
⾮公式 Texinfoフォーマット

これは sicpの⾮公式 Texinfo版訳注 1の第⼆版です。
あなたはきっと、これを Emacsの Infoモードのような Infoハイパーテキ

ストブラウザで読んでいるのでしょう。もしかすると、TEXで組版されたもの
を画⾯上で、または印刷して読んでいるのかもしれませんね。もっとも、それ
は間抜けなことだと思いますが。それに、印刷するにはだいぶお⾦がかかりま
すよね。

html-gif版は公式に無料で公開されているのですが、それを初めて Unoffi-
cial Texinfo Format(⾮公式 Texinfo版、utf)の初版として Lytha Aythが個⼈
的に変換したのは、2001年 4⽉の⻑い Emacs Lovefest Weekendの間のことで
した。

utfは html版よりも検索が簡単です。また、あまりよくないコンピュータ
(例えば寄付された 386マシンのようなもの)を使っている⼈にとってもアクセ
スしやすいでしょう。386マシンは、理論的には、Linux、Emacs、Schemeイ
ンタプリタを同時に実⾏できます。しかし、多くの 386マシンでは、Netscape
とそれに必要な X Window Systemの両⽅を動かそうとすると、まだ始めたて
でお⾦もない若いハッカーにスラッシング (thrashing)という概念を早い段階
から教えるはめになるのがオチです。それに、utf版なら圧縮なしで 1.44mb の
フロッピーディスクに収まりますので、インターネットや lanへの接続環境の
ない PCにインストールするのにも便利です。

Texinfoへの変換は、可能な範囲で直接的な翻字をしました。TEXから html
への変換のように、これによってある程度内容が損なわれてしまいました。⾮
公式 TexInfo形式では、図はいにしえの技術であるアスキーアートによって不
訳注 1英語版にはTexinfo版がありますが、⽇本語版にはありません。LATEXソースと PDF
のみです。

ix



格好に復活させられることになりました。また、多量の上付き⽂字 (‘ˆ’)と下
付き⽂字 (‘_’)を変換するうちに、曖昧さによる変換ミスが含まれてしまった
可能性も⼤いにあります。どれが変換ミスかを当てるのは、読者への課題とし
ます。しかし、少なくとも私たちは、“より⼤きい” という記号を <u>&gt;</u>
と符号化して私たちの勇敢な宇宙⾶⾏⼠たちを危険にさらすようなことはして
いません。
もしあなたが sicp.texiを変更して間違いを直したりアスキーアートを改

善したりした場合、@set utfversion 2.andresraba5.5の⾏を更新し、あなた
の差分を反映してください。例えば、もしあなたが Lythaのバージョン 1から
始めていて、あなたの名前が Bobなら、改訂版には 1.bob1, 1.bob2, …, 1.bobn
のように名前をつけるのがいいでしょう。また、utfversiondateも更新して
ください。もしあなたが⾃分の改訂版をWeb上で配布したいのなら、⽂字列
“sicp.texi”をファイルやWebページのどこかに埋め込んでおけば、Web検索
エンジンから⾒つけやすくなるでしょう。
⾮公式 Texinfo形式は、寛⼤にもフリーに配布されている html版の魂を引

き継いでいると信じられています。しかし、いつどこかの法律家の⼤艦隊が、
何でもないような⼩さなことをめぐって⼤騒ぎをしないとも限りません。です
から、あなたのフルネームを使ったり、あなたのアカウント名やマシン名を含
むかもしれない Info, dvi, PostScript, pdf版を配布したりするのは慎重に考え
てからにしてください。
Peath, Lytha Ayth

付録:Abelsonと Sussmanによる sicpのビデオレクチャーもご覧ください。
mit csail, mit ocw.

付録 2: 上記は 2001年の元の utfの紹介です。10年後のいま、utfは⼀変しま
した。数学上の記号と式は適切に組版され、図はベクターグラフィックにより
描かれています。元のテキスト形式とアスキーアートの図は今でも Texinfoの
ソースに残っていますが、表⽰されるのは Info形式にコンパイルした場合だけ
です。電⼦書籍リーダーとタブレットの夜明け時代になって、画⾯上で pdfを
読むことは、もはや⾺⿅⾺⿅しいことではないとはっきり⾔えるようになりま
した。楽しんでください！
A.R, May, 2011

⾮公式⽇本語版

SICPはかつて第⼀版、第⼆版共に⽇本にて公式に翻訳が商業出版されていま
した。第⼆版を出版していたピアソン桐原が 2013年 8⽉に ピアソングループ
から撤退し技術書の取扱を終了したため、⽇本語で SICPを読む機会は失われ
ました。このことがこの翻訳を⾏うことの契機となりました。
実際にはその後、2014年 1⽉付近に、寛⼤にも第⼆版の訳者、和⽥英⼀先

⽣がオンライン上にて SICPの訳書、「計算機プログラムの構造と解釈」全⽂
を公開して下さいました。この時点でこの⾮公式⽇本語版の価値は随分と⼩さ
くなりました。
しかし、その時、既に 3章まで翻訳していたこと、そして⾮公式 TexInfo

版が 2013年 11⽉に⼤改訂を⾏い、当初の⽇本語には正式に対応していない
texi2pdfから変更を⾏い、XeLaTeXを採⽤したために、⽇本語でも美しい組版
ができる可能性が出てきたことが、この原稿を廃棄することを押し⽌めました。

SICPのライセンスについてはインターネットアーカイブにて調べてみま
した。2001年 1⽉にMITが SICPを寛⼤にもオンラインで無料で読むことが
できるように公開された時にはライセンスが指定されていませんでした。

2008年 4⽉に MITは SICPのライセンスを CC BY-NCと指定しました。
その後ライセンスは 2011年 10⽉に⼀旦 CC BY-SAに変更されます。そして
2年後の 2013年 9⽉に再び CC BY-NCへと戻されました。この事実が SICP
原⽂のライセンスの解釈を難しくしています。ライセンスの変更はオーナーの
⾃由ですが、ライセンシーはコンテンツ取得時のライセンスを尊重すれば良い
からです。
最初に⾮公式 TexInfo版を作成した Lytha Aythはライセンス指定の無い

SICP公開をWeb⽂化に基づくものだと理解しました。次に LaTeXの組版を
開発した Andres Rabaは CC BY-SAに基き正式な許諾の下、PDF版を作成し
ました。私の翻訳は PDF版のライセンスである CC BY-SAに従うことが求め

られます。しかし、現在のMITが⾮商業を求めていることを鑑みて、Raba⽒
に許可を頂いた上で⾮商業制約を追加した cc by-nc-sa 3.0 にてリリースするこ
とにしました。

CC BY-NC、及び BY-SAは共に翻訳の許可を明記しています。従ってこの
翻訳には Lythaが⼼配したような法的問題は起こらないと信じています。しか
し同時に、法的問題は常に⼀⽅的に起こされることがあることもまた現実です。
従って読者の皆様には常にネットワーク上のデータは (そしてプログラムも!)
消えてなくなってしまうシャボン⽟であることを忘れずに御⽤⼼願います。

TeX、LaTeX 環境の⽇本語対応を進めて下さった全ての関係者の皆様に
感謝します。特に最新の情報を常に更新し続けて下さっているTeX Wikiの奥
村 晴彦⽒、W32TeXを⾃動でインストールし更新可能な TeXインストーラ作
者の阿部 紀⾏⽒、XeLaTeX向け⽇本語パッケージ “ZXjatype” を開発して下
さった⼋登 崇之⽒に感謝致します。
海外では SICPの新しい形の開発が⾮常に盛んです。PDFはもちろん、epub

やインタラクティブ版、Kindle版 (mobi形式)、Clojureや JavaScriptによる
SICP等が公開されています。この翻訳は CC BY-NC-SAですので⾮商業であ
ればそのような派⽣や翻案に利⽤することが可能です。⽇本でも SICPの世界
が広がっていくことを期待しています。

※ 校正御協⼒者様 (順不同、敬称略)

• Kei Shiratsuchi
• Kimura, Koichi
• のな
• Naoki Ainoya

xii

http://creativecommons.org/licenses/by-nc-sa/3.0/
http://oku.edu.mie-u.ac.jp/~okumura/texwiki/
http://www.math.sci.hokudai.ac.jp/~abenori/soft/abtexinst.html
http://zrbabbler.sp.land.to/zxjatype.html
https://github.com/kei-s
https://github.com/kimurakoichi
https://github.com/nna774
https://github.com/ainoya


⾮公式⽇本語版翻訳改訂版

この翻訳改訂版は、minghai⽒の⾮公式⽇本語版 (以降、minghai⽒版)の翻訳
に不⼗分なところが多かったため、原著から翻訳をし直したものです。この翻
訳を進めるにあたっては、minghai⽒版の訳を置き換えていくというやり⽅で
進めていきました。しかし、差分を取ればわかっていただけると思いますが、
minghai⽒版のテキストは痕跡をとどめていないはずです。この⽅式を採った
のは、主にモチベーション上の理由によるものです。もしminghai⽒版の痕跡
が (誰が訳しても同じようになるものを除いて)あるとすれば、それは作業漏れ
が残っているということですので、ご報告いただければ幸いです。
⽤語については、無⽤の混乱を避けるため、和⽥英⼀⽒によるオンライン

版「計算機プログラムの構造と解釈」(以下、和⽥⽒版)にできるだけ追従する
ようにしました。もちろん、訳⽂はすべて⾃分で⼀から作り上げたものです。
⽤語の追従によって著作権やライセンス上の問題が発⽣することはないと信じ
ていますが、問題があるようであればお教えいただければと思います。
私としては、もし minghai⽒版がなければ、和⽥⽒版の訳書を購⼊して勉

強して、それで済ませていたはずのところです。訳⽂は硬いとはいえ読めない
ことはなく、英語で読むよりは速く進められそうでした。ほかのこなれていな
い訳書同様、不満を持ちながら読み、それで終わりだったはずです。
しかし、minghai⽒版が出てきたことで、状況は⼤きく変わりました。minghai

⽒版は無料の PDFなので、有料の書籍とオンライン HTMLという和⽥⽒版よ
りアクセスが容易です。さらに、和⽥⽒版は (Amazonのレビューにもあるよ
うに)翻訳の評判がよくありませんが、minghai⽒版はですます調で⼀⾒親し
みやすく、翻訳に問題が多いということがすぐにはわかりません。
これでは、SICPの評判を聞いて勉強しようとした⼈が、minghai⽒版に流

れてしまうのではないか。私のその危惧は、Twitterのタイムラインで現実化し
ていました。タイムラインでもminghai⽒版に⼿を伸ばそうとしている⼈がい

たため、和⽥⽒版をお勧めしておきました。もちろん、その⼈は氷⼭の⼀⾓で、
実際には少なからぬ⼈たちがminghai⽒版に流れたのではないかと思います。
⽂体が硬くて意味の取りにくい翻訳の場合、読者は何とか読み解く努⼒を

するか、読むことをあきらめて訳者を責めるか (これは Amazonレビューで起
こっていることです)のどちらかの⾏動を取るでしょう。後者の⼈に対しては
(有料で購⼊した場合)⾦銭的な損害を与えているかもしれませんが、それがす
べてです。どうしても読みにくければ、原⽂で読むことになるでしょう。いず
れにせよ、翻訳が存在しなかった場合以上に悪いことにはなりません。
しかし、表⾯上読みやすそうな問題の多い翻訳は、はるかに⼤きな副作⽤

をもたらす可能性があります。実際には誤訳のために訳⽂が意味不明になって
いるだけなのに、理解できないのは書いてある内容が難しすぎるためだ、それ
を理解するだけの頭脳が⾃分にないためだ、と読者は思い込んでしまうかもし
れません。それだけはどうしても避けたいところです。
それらのことを考え合わせた結果、SICP を原⽂から翻訳しながら勉強し

ていくことにしました。始めたとき、毎⽇⼀時間というペースで進めていくと
決め、数⽇間やってみた進捗状況から翻訳が終わるまでの期間を⾒積もったと
ころ、どうやら⼀年ぐらいかかりそうだとわかりました。⼀年というのは⻑い
時間です。⼼がくじけそうになりつつも、ミヒャエル・エンデの「モモ」を思
い出し、その⽇その⽇の分だけを考え、続けていきました。毎⽇夜 9時からを
定刻としていたため、Twitterでは「時報」と⾔ってくれる⼈もいました。
そして先⽇、ようやく翻訳とチェック作業が終わりました。期間は予定通

り⼀年程度でした。勉強を兼ねての翻訳だったため、中⾝を理解しながら、練
習問題もできるだけ (といっても、第 3章の途中あたりでリタイアしましたが)
解いていきました。前評判に違わず、SICPの内容はとても充実していました。
翻訳前に危惧していたような、「この時間で何ができたか」と後悔するような
ことにはならずにすみました。
もちろん、私のこの翻訳が完全だとは思っていません。しかし、致命的な

誤訳が何百箇所もあってどこから⼿をつけていいかもわからないような訳⽂で
ないことは保証します。この本の内容が理解できないとすれば、それは翻訳が
おかしいからではありません。もちろん、内容が難しすぎるためでも、それを
理解するだけの頭脳がないためでもありません。⾜りないのは時間をかけた読
み込みです。じっくりと⼀⽂⼀⽂の意味を考え、サンプルコードを動かし、で
きるだけ練習問題を解いていけば、それで⼗分です。この本は、読者を置き去
りにしようとするような本ではありません。
最後に、⽇本語版のテンプレートを提供していただいたことについて、

minghai⽒に深く感謝します。minghai⽒の⽇本語版テンプレートがなければ、

私の TEXや LATEXについての限られた知識では⽇本語訳を⾏うことはできな
かったでしょう。
それでは、SICP⽇本語版をお楽しみください。
真鍋宏史

献辞

この本を、尊敬と賛美を込めて、コンピュータの中に住む妖精に捧げます。
“コンピュータサイエンスに関わる私たちにとって、コンピューテ
ィングを楽しむことはとても⼤事だと私は思っています。コンピ
ューティングの始まりの時期には、それはとても楽しいことでし
た。ご存じの通り、お⾦を払うお客様たちが騙されるということ
がそこかしこで起こり、それからしばらくして、私たちは彼らの
不満を真⾯⽬に受け取るようになりました。私たちは、あたかも
コンピュータをきちんとエラーなしで完璧に使えるようにするこ
とが私たちの責任であるかのように考え始めてしまいました。私
はそうは思いません。私たちは、コンピュータを発展させ、新し
い⽅向に向かわせ、家の中で楽しめるようにしておくことに責任
があると思います。コンピュータサイエンスのその分野が、ずっ
とその楽しさの感覚を失わないでくれればと願っています。特に、
私たちが伝道師になったりしないことを望んでいます。聖書のセ
ールスマンのような感覚を持たないでください。そんな⼈たちは、
もう世界に⼗分すぎるほどいます。あなたがコンピューティング
について知っていることは、ほかの⼈たちも学ぶことができます。
コンピューティングをうまくやる鍵が、あなたの⼿の中にしかな
いとは思わないでください。あなたの⼿の中にあるものは知性だ
と、私は思い、またそうであることを願っています。それはつま
り、あなたが出会ったマシンに対して、出会ったときよりも多く
のものを⾒ることができる能⼒です。そして、それをそれ以上の
ものにできる能⼒です。”
—Alan J. Perlis (April 1, 1922 – February 7, 1990)

前書き

教育者、将軍、栄養⼠、精神分析医、そして両親は、計画を⽴て (program)ま
す。軍隊や学⽣は計画によって動いていますし、計画によって動いている社会
もあります。⼤きな課題に取り組むときにはいくつもの計画を使いますが、そ
れらのほとんどは問題に取り組む中で⽣まれるものです。それらの計画には、
そのとき取り組んでいる課題に特化しているように⾒える問題があふれていま
す。計画を⽴てるということ⾃体を知的な活動として理解するためには、コン
ピュータプログラミングに⽬を向ける必要があります。コンピュータプログラ
ムを読み書きする必要があります— それも、⼤量に。何のプログラムか、どう
いう応⽤のためかというところはあまり重要ではありません。重要なのは、そ
れらがどのようにふるまい、またより⼤きなプログラムを作る中でそれらがど
れだけスムーズにほかのプログラムと協調できるかということです。プログラ
マは、部品の完全性と、組み合わせたものの妥当性の、両⽅を追求しなければ
いけません。この本では、“プログラム” というのは、Lispのある⽅⾔で書かれ
た、デジタルコンピュータ上で実⾏されることを前提としたプログラムの作成
・実⾏・研究に焦点を当てています。Lispを使うことは、プログラムの記述に
ついて制約を加えるだけで、何をプログラムするかを制約するものではありま
せん。
この本の主題について考えるうえで、3つの事象を考えることになります。

⼈間の⼼、コンピュータプログラムの集合、そしてコンピュータです。コンピ
ュータプログラムはすべて、実際のプロセスや精神的なプロセスのモデルであ
り、⼈間の⼼の中で⽣まれるものです。これらのプロセスは、⼈間の経験と思
考から浮かび上がってくるもので、数はとても多く、細かいところまで⼊り組
んでいて、いつでも部分的に理解することしかできません。それらがコンピュ
ータプログラムとして永遠に満⾜できるようにモデル化されるなんていうこと
は、めったにありません。ですから、私たちのプログラムは、注意深く⼯夫さ

れたばらばらの記号の集合であり、絡まり合う関数の組み合わせでありながら
も、絶えず進化していきます。私たちのモデルに対する理解が深まったり、広
がったり、⼀般化したりするたびに、私たちはプログラムに⼿を加えていきま
す。最終的に、モデルは準安定状態に達し、それは私たちが取り組む別のモデ
ルの中に位置を占めることになります。コンピュータプログラミングに関する
興奮の源は、プログラムとして表現されたメカニズムが私たちの⼼の中とコン
ピュータ上で絶え間なく広がっていくことと、またそれが⽣み出す知覚の爆発
です。芸術が私たちの夢を解釈するものだとすると、コンピュータは夢を、プ
ログラムという形で実⾏するものなのです！
コンピュータは強⼒ですが、厳しい親⽅でもあります。プログラムは正し

くなければならず、伝えたいことは細かいところまで正確に伝えなければいけ
ません。ほかのすべての記号的な活動と同じく、プログラムが真であることは
論証によって知ることができます。Lispはそれ⾃⾝、意味論 (これもまたモデ
ルです)を割り当てることができ、そしてプログラムの仕様が、例えば述語論理
などによって決められるのであれば、論理学の証明⽅法によって、容認可能な
正確性の論証をすることができます。残念ながら、プログラムが⼤きく複雑に
なるにつれ (いつもそうなるのですが)、仕様⾃体の妥当性、⼀貫性、正しさが
疑わしいものになっていきます。そのため、完全な形式的な正しさの論証とい
うものが⼤きなプログラムについてくるということはめったにありません。⼤
きなプログラムは⼩さなものから成⻑していくものなので、正しさについて確
信を持てるような標準的なプログラム構造 (これをイディオムと呼びます) の
武器庫を作り上げ、確⽴された組織化技術によってより⼤きな構造に組み⽴て
ていく⽅法を学ぶことが決定的に重要になります。この本では、それらの技術
について詳細に扱っていきます。それらを理解することは、プロメテウスのよ
うな創造的な営みであるプログラミングという分野に加わるためには不可⽋な
ことです。何よりも、強⼒な組織化の技術を発⾒し⾝につけていくことによっ
て、⼤規模で重要なプログラムを書くにあたって必要となる能⼒を育てること
ができます。また逆に、⼤規模なプログラムを書くことは負担がとても⼤きい
ということもあり、何とか新しい⼿法を発明して、⼤規模なプログラムに組み
込む関数や実装の量を減らすことができないかと考えていくことになります。
プログラムとは違って、コンピュータは物理法則に従わなければなりませ

ん。もし素早く—状態遷移あたり数ナノ秒といったレベルで—動作したけれ
ば、電⼦をほんの短い距離 (せいぜい 1 1

2
フィートだけ)しか運ぶことができま

せん。空間的にぎっしり詰め込まれた膨⼤な数のデバイスが発する熱は、取り
除く必要があります。⾼機能性とデバイス密度のバランスを取りながら、精緻
な⼯学的技法が発展してきました。いずれにせよ、ハードウェアは私たちがプ

ログラムしようとしているレベルよりも基礎的なレベルで動いています。私た
ちの Lispのプログラムを “機械の”プログラムに変換する処理は、それ⾃⾝が
抽象的なモデルで、それは私たちがプログラムするものです。その研究と作成
は、どんなモデルをプログラムするときにも関わってくるような組織的プログ
ラムについて、多くの洞察を与えてくれます。もちろん、コンピュータそれ⾃
⾝もそのようにモデル化されています。考えてみてください。⼀番⼩さい物理
的なスイッチング素⼦は量⼦⼒学によってモデル化され、量⼦⼒学は微分⽅程
式によって記述され、微分⽅程式の詳細なふるまいは数値近似によって表され、
数値近似はコンピュータプログラムによって表現され、そのコンピュータプロ
グラムはコンピュータによって実⾏され、そのコンピュータを構成する素⼦は
. . .！

3つの焦点を別々に設定することは、単なる⽅法上の便利さのためだけで
はありません。よく「すべては頭の中のことだ」と⾔われますが、このように
論理的に分けることによって、これらの焦点の間の記号的通信を促進できます。
⼈間の経験の中で、これら 3つの豊かさ、⽣命⼒、潜在⼒を超えるものは、⼈
⽣そのものの発展ぐらいです。これらの焦点の間の関係は、せいぜい準安定状
態です。コンピュータの容量の⼤きさや速さは、これで⼗分となることは決し
てありません。ハードウェア技術の進歩のたびに、より⼤きなプログラミング
の取り組み、新しい組織化原則、そして抽象モデルの発展が引き起こされます。
読者は、定期的に “どこに向かっているんだ? どこに向かっているんだ?”と⾃
問しなければなりません。—でも、あまり問いすぎないようにしましょう。そ
うでないと、ほろ苦い哲学的な便秘のために、プログラミングの楽しさを⾒逃
してしまうことになります。
私たちが書くプログラムの中で、⼀部のもの (しかし、決して⼗分にはなら

ない)は、並べ替え、数列の最⼤値を⾒つける、素数性判定、平⽅根の計算と
いった、精密な数学的な関数を実⾏します。そのようなプログラムはアルゴリ
ズムと呼ばれ、その最適なふるまいについては多くのこと (特に、実⾏時間と
データストレージの必要量という⼆つのパラメータについて)が知られていま
す。プログラマは、よいアルゴリズムとイディオムを⾝につけなければなりま
せん。プログラムの中には厳密な仕様が記述しにくいものもありますが、プロ
グラマとしては責任を持ってパフォーマンスを推定し、また常に改善しようと
しなければなりません。

Lispは、“⽣き残ったもの”です。約四半世紀にわたって使われてきていま
す。活発なプログラミング⾔語の中で、これより⻑い時間を⽣きてきたものは
Fortranだけです。どちらの⾔語も、重要な応⽤領域でのプログラミングの必
要性を⽀えてきています。Fortranは科学計算と⼯学計算を⽀え、Lispは⼈⼯

知能を⽀えてきました。この⼆つの分野は今でも重要であり続け、これらの分
野のプログラマは Lispと Fortranという⼆つの⾔語に⼒を注いでいるので、少
なくとももうあと四半世紀の間活発に使われる⾔語であり続けるということは
⼗分ありうることです。

Lispは変化します。このテキストで使っている Scheme⽅⾔は、オリジナ
ルの Lispから進化してきたもので、いくつかの重要な点で異なっています。例
えば、変数束縛のための静的スコーピングや、関数が関数を値として返すこと
を許可することなどです。構⽂的な構造において、Schemeは Algol 60に似て
います。その近さは、Schemeと初期の Lispとの近さと同じ程度です。Algol
60は、もう活発な⾔語になることはないでしょうが、Schemeと Pascalの遺
伝⼦の中に⽣き続けています。これら⼆つの⾔語ほど、かけ離れた⽂化に取り
囲まれ、その流通貨幣となっているような⼆つの⾔語というのもなかなかない
でしょう。Pascalはピラミッドを建てるためのものです。雄⼤で、息をのむよ
うな、静的な構造物で、軍隊が重い⽯を運び、所定の位置に押しこむことによ
って作られています。Lispは有機体を作るためのものです。雄⼤で、息をのむ
ような、動的な構造物で、⼩さなチームが無数のより単純な揺れ動く有機物を
所定の位置に⼊れることによって作られています。どちらの場合にも、組織化
のための原則は同じですが、ひとつだけ⾮常に重要な違いがあります。個々の
Lispプログラマに委ねられたエクスポート可能な関数は、Pascal企業の中で⾒
つけられるものより桁違いに多いのです。Lispのプログラムは、本来⽬的とし
ていたアプリケーションを超えた有⽤性を持つ関数でライブラリをいっぱいに
します。そのように有⽤性が増加していく背景には、リストという Lispのネイ
ティブデータ構造の存在があります。リストの単純な構造と⾃然な適⽤可能性
は、驚くほど普遍的な関数という形で表れてきます。Pascalでは、宣⾔可能な
データ構造の多さが関数の専⾨化を引き起こし、気軽な連携は禁じられ、ペナ
ルティを与えられるものとなります。ひとつのデータ構造を扱う 100の関数が
あるほうが、10のデータ構造を扱う 10の関数があるのよりもよいのです。結
果として、ピラミッドは千年もの間変化せずに存在し続けなければならないの
に対し、有機体は存在する限り進化し続けなければなりません。進化を⽌める
ということは滅びることを意味します。
この違いをはっきり知るためには、この本の中での教材と練習問題の扱い

を、何でもいいので Pascalを使った⼊⾨テキストと⽐べてみてください。この
テキストは mitでしか消化できない、mitに固有の種族のためのものだとかい
った妄想は抱かないでください。このテキストは、真⾯⽬な Lispのプログラミ
ングの本はどのようなものでなければいけないかというところを形にしたもの
であり、学⽣が誰であるか、どこで使われるかは関係ありません。

ひとつ気をつけておいていただきたいのは、この本はプログラミングに関
するテキストだということです。ほとんどの Lispの本は⼈⼯知能に使うため
の準備となっていますが、この本は違います。いずれにしても、対象となるシ
ステムが⼤きくなるにつれ、ソフトウェア⼯学と⼈⼯知能に関わる重要なプロ
グラミング上の課題は融合していく傾向にあります。まさにそのために、⼈⼯
知能以外の領域でも、Lispに対してこれほどの関⼼が向けられているのです。
その⽬的とするところから予想できる通り、⼈⼯知能の研究は数多くの重

要なプログラミングの問題を⽣み出しています。ほかのプログラミング分野で
あれば、このような⼤量の問題に対しては新しい⾔語が⽣み出されるところで
す。実際に、⾮常に⼤きなプログラミングのタスクにおいて、組織化のために
便利な⽅法のひとつは、新しい⾔語を発明して、タスクモジュールの中に通信
を閉じ込め、コントロールするということです。これらの⾔語は、私たち⼈間
が頻繁にやりとりを⾏うシステムの周辺部に近づくにつれて、単純でなくなっ
ていく傾向があります。その結果、そのようなシステムでは、複雑な⾔語処理
の機能が何回も繰り返し実装されることになります。Lispは、⾮常に単純なシ
ンタックスとセマンティクスを持つため、パージングは初歩的なタスクとして
扱うことができます。ですので、パージング技術は Lispのプログラムの中でほ
とんど何の役割も果たしません。また、⼤きな Lispのシステムが成⻑し変化し
ていく中で、⾔語処理系の構築がその成⻑速度の妨げになることはめったにあ
りません。最後になりますが、このシンタックスとセマンティクスの単純さこ
そが、すべての Lispプログラマが付き合うことになる重荷と⾃由の源泉にな
っているのです。数⾏以上の Lispプログラムを書こうとすると、必ず⾃由に使
える関数でいっぱいになります。発明し、組み込み (fit)ましょう。かんしゃく
を起こし (have fits)、再発明しましょう！括弧のネストの中に⾃分の考えを書
いていく Lispプログラマに乾杯。
Alan J. Perlis
New Haven, Connecticut

第⼆版 序⽂

ソフトウェアが、ほかの何にも似ていないということはあるでし
ょうか。それは捨てられるためのもので、常にシャボン⽟のよう
に⾒ることがすべてだと。
—Alan J. Perlis

この本で扱われている題材は、1980年から mitの⼊⾨レベルの計算機科学科⽬
の基礎となってきました。初版が発売されたとき、すでにこの教材を使った授
業を 4年間⾏っていました。この第⼆版が出てくるまでには、それからさらに
12年が過ぎています。私たちの成果が広く受け⼊れられ、ほかのテキストにも
取り込まれていることについて、私たちはうれしく思っています。私たちは、
学⽣たちがこの本に出てくる考え⽅やプログラムを⾝につけ、それらを新しい
コンピュータシステムやコンピュータ⾔語の核として組み込むのを⾒てきまし
た。古代のタルムードに出てくる⾔葉遊びを⽂字通り実現し、私たちの学⽣は
建築者になったのです。このような有能な学⽣、功績ある建築者を得ることが
できたのは幸運なことでした。
この版の準備にあたって、何百にものぼるわかりやすい説明を取り込みま

した。それらは、私たち⾃⾝の教育経験や、mitやほかの場所の同僚たちのコ
メントから得られたものです。この本に出てくる主なプログラミングシステム
のほとんどを再設計しました。それには⼀般計算システム、インタプリタ、レ
ジスタマシンシミュレータ、コンパイラが含まれます。さらに、プログラムの
例をすべて書き換え、ieee Scheme標準 (IEEE 1990)に適合した Scheme実装
であればどんなものの上でもコードを実⾏できるようにしました。
この版では、いくつかの新しいテーマに重点を置いています。これらの中

で最も重要なものは、計算モデルの中で時間を扱うさまざまなアプローチの中
⼼的な役割を果たしているものです。それらのアプローチには、状態つきのオ

ブジェクト、並⾏プログラミング、関数型プログラミング、遅延評価、⾮決定
性プログラミングがあります。並⾏性と⾮決定性については新たに節を設け、
本全体を通してこのテーマを統合させるよう⼒を尽くしました。
この本の初版は、mitの半期の科⽬のシラバスをなぞったものでした。第⼆

版では、多くの新しい題材を含めたため、半期ですべてをカバーするのは無理
でしょう。ですので、講師は教えるべき内容を選ぶ必要があります。私たち⾃
⾝が講義をする際には、時々論理プログラミング (4.4節)を⾶ばします。学⽣
にはレジスタマシンのシミュレータを使わせますが、その実装 (5.2節)はやり
ません。そして、コンパイラ (5.5節) はざっと概要を⾒るだけです。それでも
なお、これは中⾝の濃い授業です。講師によっては、最初の 3章か 4章だけを
やって、残りの題材は次の科⽬に譲りたいという⼈もいるでしょう。

World-Wide-Webサイト http://mitpress.mit.edu/sicp は、この本の読者に
対するサポートを提供しています。それには、この本に出てくるプログラム、
プログラミング練習問題のサンプル、補助教材、ダウンロード可能な Lispの
Scheme⽅⾔の実装が含まれます。

第⼀版 序⽂

コンピュータはバイオリンのようなものです。次のような場⾯を
想像してみてください。ある初⼼者の⼈が、まず蓄⾳機を動かし
てみて、それから次にバイオリンを触ってみます。その⼈は、バイ
オリンの⾳はひどいと⾔います。⼈間主義者たちや多くの計算機
科学者から私たちが聞く話はそういったものです。コンピュータ
のプログラムは特定の⽬的にはよいが、柔軟ではないと。バイオ
リンやタイプライターだってそんなものです。使い⽅を学ばなけ
ればの話ですが。
—Marvin Minsky, “Why Programming Is a Good Medium for Ex-
pressing Poorly-Understood and Sloppily-Formulated Ideas”

“The Structure and Interpretation of Computer Programs”(SICP, 計算機プロ
グラムの構造と解釈)はマサチューセッツ⼯科⼤学 (MIT)での⼊⾨者レベルの
計算機科学の科⽬です。mit で電気⼯学や計算機⼯学を専攻する学⽣はすべて、
4つある “共通コアカリキュラム”のひとつとして履修しなければなりません。
残りは、回路と線形システムについての科⽬ 2つと、デジタルシステム設計の
科⽬ひとつです。私たちは、この科⽬の開発に 1978年から携わってきました。
そして、1980年の秋からは、この教材の現在の形で毎年 600⼈から 700⼈の学
⽣を教えてきています。ほとんどの学⽣はコンピュータについてそれまでまっ
たく秩序⽴ったトレーニングを受けていないか、受けていてもわずかなもので
した。とはいえ、コンピュータでちょっと遊んだことのある学⽣は多くいまし
たし、プログラミングやハードウェア設計に相当な経験のある学⽣も数⼈はい
ましたが。
私たちがこの計算機科学の⼊⾨科⽬の授業計画を策定するにあたって、⼆

つの懸念点を内容に反映させました。⼀つ⽬は、コンピュータ⾔語はただコン

ピュータに操作を実⾏させる⽅法であるにとどまらず、⽅法論についての考え
⽅を表現するための今までにない形式的媒体であるという考え⽅を打ち⽴てた
いということでした。ですから、プログラムは⼈間が読めるように書かなくて
はならず、それがたまたま機械によって実⾏できるというだけのことです。⼆
つ⽬は、私たちがこのレベルの科⽬で扱われるべき中⼼的な題材について信じ
ていることです。教えるべきはことは、特定のプログラミング⾔語の⽂法では
なく、また特定の関数を計算するための賢いアルゴリズムでもなく、さらに⾔
うとアルゴリズムの数学的解析やコンピューティングの基礎ですらなく、重要
なのは⼤きなソフトウェアシステムの知的複雑性をコントロールする技術であ
るということです。
私たちの⽬標は、この科⽬を修了した学⽣たちが、プログラミングのスタ

イルと美学について、基礎的な感覚を⼗分に⾝につけてくれることです。学⽣
たちが、⼤きなシステムの複雑性をコントロールするための主要な技術を使い
こなせるようになること。模範的なスタイルで書かれていれば、50ページにわ
たるプログラムでも読めるようになっていること。どんなときにも、何を読む
べきでないか、何を理解しないでもいいかを判断できるようになること。また、
プログラムを変更するにあたって、⾃信を持って、元の作者の意図とスタイル
を保ったままで変更を加えることができるようになっていること。
これらのスキルは、決してコンピュータプログラミングに特有のものでは

ありません。私たちが教え、また利⽤する技術は、すべての⼯学的設計に共通
するものです。私たちは、詳細な部分を隠す抽象化を適宜加えることによって、
複雑性をコントロールします。私たちは、共通のインターフェイスを確⽴し、
標準的かつ中⾝のよくわかっている部品を「うまく組み合わせる」やり⽅で、
複雑性をコントロールします。私たちは、設計を記述するためにさまざまな⾔
語を作り出し、それぞれの⾔語が設計のある特定の側⾯を際⽴たせ、それ以外
の側⾯を⽬⽴たなくさせることによって、複雑性をコントロールします。
私たちのこの科⽬に対するアプローチの底にあるものは、“計算機科学”は

科学ではなく、その意義はコンピュータとほとんど関係がないという信念です。
コンピュータ⾰命とは、私たちの考え⽅と、考えを表現するやり⽅の⾰命です。
この変化を象徴する出来事は、⼿続き的認識論 (procedural epistemology)—古
典的な数学的テーマで取られるような宣⾔的な⾒⽅ではなく、命令的な⾒⽅に
よる知識構造の研究—の出現でしょう。数学は、“何であるか”という考えにつ
いて正確に扱う枠組みを提供します。計算機科学は、“どうやるか”という考え
について正確に扱う枠組みを提供します。
この教材では、プログラミング⾔語 Lispのひとつの⽅⾔を使います。⾔語

を正式に教えることはしません。その必要がないからです。普通に使っていく

だけで、数⽇のうちに学⽣は⾝につけます。これは Lisp系の⾔語の⼤きな利
点です。複合式を作るやり⽅は少ししかなく、構⽂構造はあってないようなも
のです。形式的な性質はすべて、チェスのルールのように、⼀時間もあればカ
バーできます。少ししたら、⾔語の構⽂的な詳細について忘れてしまい (元か
らそんなものはないので)、実際の問題— 何を計算したいのか、どうやって問
題を扱えるサイズの部品に分解するか、それらの部品をどのように作っていく
か—に取り組むことになります。もうひとつの Lispの利点は、私たちの知る限
りほかのどんな⾔語よりも、プログラムをモジュールに分解する⼤規模な戦略
をサポートしてくれる (しかし強制はしない)というところです。⼿続きやデー
タによる抽象化をしたり、よくある使い⽅のパターンを捉えるために⾼階関数
を使ったり、代⼊とデータ変更によって局所状態をモデル化したり、ストリー
ムと遅延評価によってプログラムの部品同⼠を連携させたりといったことが可
能です。さらに、組み込み⾔語の実装も簡単にできます。また、これらすべて
は、漸進的なプログラム設計、構築、テスト、デバッグに対する優れたサポー
トを提供する対話的環境に埋め込まれています。これまでにないパワーと優美
さを備えたすばらしいツールを作り上げた John McCarthyをはじめ、これま
でのすべての世代の Lispウィザードたちに感謝の気持ちを捧げます。
私たちが使う Lispの⽅⾔である Schemeは、Lispと Algolのパワーと優美

さをひとつにまとめようとしたものです。Lispからは、シンプルな構⽂によ
るメタ⾔語的なパワー、プログラムをデータオブジェクトとして表す統⼀的な
やり⽅、ガベージコレクションつきヒープへのデータ割り当てが取り込まれて
います。Algolからは、レキシカルスコープとブロック構造を取り込んでいま
す。これらは、Algol委員会のプログラミング⾔語設計の先駆者たちからの贈
り物です。チャーチの λ計算とプログラミング⾔語の構造との間の関係につい
ての洞察は John Reynoldsと Peter Landinによるものです。また、コンピュ
ータが世界に現れる何⼗年も前からこの領域を探索してきた数学者たちに対す
る恩義も感じています。彼ら開拓者たちには、Alonzo Church, Barkley Rosser,
Stephen Kleene, Haskell Curryなどが含まれます。

xxvi



謝辞

この本とこのカリキュラムの開発を⼿助けしてくださった多くの⽅々に感謝い
たします。
私たちの科⽬は、明らかに “6.231”の知的な系譜を継ぐものです。それは、

1960年代後半に mitで Jack Wozencraftと Arthur Evans, Jr.が担当していた、
プログラミング⾔語学と λ計算についてのすばらしい科⽬でした。

Robert Fanoが mitの電気⼯学と計算機科学の⼊⾨カリキュラムを再編成
し、⼯学設計の原理に重点を置くようにしてくれたことに⾮常に感謝していま
す。彼はこの取り組みを始めるよう私たちを導き、最初の科⽬ノートをまとめ
てくれました。この本は、それをもとに発展してきたものです。
私たちが教えようとしているプログラミングのスタイルと美学は、その

多くが Guy Lewis Steele Jr.と連携して開発したものです。彼は、Gerald Jay
Sussmanとともに Scheme⾔語の初期の開発に携わりました。さらに、David
Turner, Peter Henderson, Dan Friedman, David Wise, Will Clingerは、この本
に登場する関数プログラミングコミュニティのテクニックを数多く教えてくれ
ました。

Joel Mosesは、⼤きなシステムを構築する⽅法を教えてくれました。彼は、
記号計算に使われる Macsymaシステムでの開発経験を通して、制御の複雑性
を回避して、モデル化しようとしている世界の現実の構造を反映するようデー
タを体系化することに集中することが重要だということを学んでいました。

Marvin Minskyと Seymour Papertは、プログラミングと、知的⽣活の中
でそれをどのように位置づけるかについて、私たちの姿勢の多くを形作ってく
れました。コンピュータがなければ正確に扱うには複雑すぎるような考えを探
究するにあたって、コンピュータが表現⼿段を与えてくれるということを理解
できるようになったのは彼らのおかげです。彼らは、学⽣のプログラムを書き、
変更する能⼒が、⾃然活動を探究するための強⼒な⽅法になると強調してい

xxvii



ます。
私たちはまた、プログラミングはとても楽しいものであり、プログラミン

グの楽しみを⽀えるよう注意しなければいけないという Alan Perlisの考え⽅
に強く同意します。この楽しみの⼀部分は、偉⼤な職⼈たちの働いているとこ
ろからも得ることができます。Bill Gosperや Richard Greenblattのもとで⾒
習いプログラマとして働けたことは幸運なことでした。
私たちのカリキュラムの開発に貢献してくださったすべての⽅々について

名前を挙げるのは困難です。過去 15年間私たちと共に働き、私たちの科⽬に
多くの時間を費やしていただいたすべての講師、⼝頭の指導者、チューターた
ち、中でも Bill Siebert, Albert Meyer, Joe Stoy, Randy Davis, Louis Braida,
Eric Grimson, Rod Brooks, Lynn Stein, Peter Szolovitsに感謝します。現在は
ウェルズリーにいる Franklyn Turbakに、教育上の際⽴った貢献について特に
感謝します。彼の学部⽣向けの授業は、私たち皆が⽬指す基準を打ち⽴てまし
た。Jerry Saltzerと Jim Millerには、私たちが並⾏性のミステリーに取り組む
のを⼿助けしてくださったことを感謝します。そして Peter Szolovitsと David
McAllester には、第 4 章における⾮決定性評価の説明に対する貢献に感謝し
ます。
多くの⽅々が、他⼤学でこの資料を紹介するのに⼤きな努⼒を費やしてく

ださいました。中でも私たちが緊密に連携していたのは、イスラエル⼯科⼤学
の Jacob Katzenelson、カリフォルニア⼤学アーバイン校の Hardy Mayer、オ
ックスフォード⼤学の Joe Stoy、パデュー⼤学の Elisha Sacks、ノルウェー技
術科学⼤学の Jan Komorowskiです。私たちは、他⼤学にこの科⽬を適合させ
ることで主要な教育の賞を受けた同僚たちを⾮常に誇りに思います。この中に
は、イェール⼤学の Kenneth Yip、カリフォルニア⼤学バークレー校の Brian
Harvey、コーネル⼤学の Dan Huttenlocherが含まれます。

Al Moyéは、ヒューレット・パッカードの技術者にこの教材を教え、その
講義をビデオテープにすることを⼿配してくれました。これらのビデオを使っ
た⽣涯教育のコースを計画し、世界中の⼤学や企業で教えた有能な講師たち、
特に Jim Miller, Bill Siebert, Mike Eisenbergに感謝します。
第⼀版の翻訳には、ほかの国の多くの教育者が多⼤な労⼒をつぎ込んでく

れました。フランス語版はMichel Briand, Pierre Chamard, André Pic、ドイ
ツ語版は Susanne Daniels-Herold、⽇本語版は元吉⽂男⽒によって製作されて
います。中国語版は誰が製作したのかわかりませんが、“無許可”翻訳の題材と

1
⼿続きを⽤いた抽象化の構築

⼼が単純な考えについてその⼒を発揮する活動は、主に次の 3つ
です。1. 単純な考えをひとつの合成物に組み合わせること。すべ
ての複雑な考えはこれによってできています。2. ⼆番⽬は、単純
なものでも複雑なものでも、⼆つの考えをひとつの場所に並べて
置き、それらをひとつに結合することなしに同時に眺めることで
す。これによって、⼼は関係性についての考えを得ることができま
す。3. 三番⽬は、その考えの実在に伴うほかのすべての考えから、
その考えを切り離すことです。これは抽象化と呼ばれ、これによ
ってすべての⼀般概念というものはできています。—John Locke,
An Essay Concerning Human Understanding (1690)

私たちは、計算プロセス (computational process)について学ぼうとしています。
計算プロセスというのは抽象的な存在で、コンピュータの中に⽣息しています。
それらが進化するにつれ、プロセスはデータ (data)と呼ばれるまた別の抽象物
を操作します。プロセスの進化は、プログラム (program) と呼ばれる規則のパ
ターンによって指揮されています。プログラムを作るのは、プロセスを指揮す
るためです。事実上、私たちは呪⽂を使ってコンピュータの精霊に魔法をかけ

ているのです。
計算プロセスは、魔法使いの⾔葉で⾔うと、精霊のようなものです。それ

は、⾒たり触れたりすることはできません。そもそも、物質によって作られて
いないのです。それでも、それはとてもリアルな存在です。計算プロセスは、
知的な作業を⾏うことができます。質問に答えることができます。銀⾏のお⾦
を⽀出したり、⼯場でロボットの腕を操作したりすることで、世界に影響を与
えることもできます。私たちがプロセスに魔法をかけるのに使うプログラムと
は、魔法使いの呪⽂のようなものです。それは難解で深遠なプログラミング⾔
語 (programming language)による記号的表現によって慎重に組み⽴てられ、プ
ロセスに実⾏させたいタスクを指⽰します。
計算プロセスは、正しく動くコンピュータでは、プログラムを正確かつ精

密に実⾏します。そのため、魔法使いの⾒習いのように、初⼼者プログラマは
魔法のもたらす結果を理解し、予測することを学ばなければいけません。プロ
グラムのほんの⼩さな間違い (普通、バグ (bug)やグリッチ (glitch)と呼ばれま
す)も、複雑な予期しない結果を引き起こすことがあるのです。
運のいいことに、プログラムを学ぶことは魔法を学ぶことに⽐べるとずっ

と安全です。私たちが扱う精霊は、都合のいいことに安全な⽅法で封じ込まれ
ているからです。しかし、現実世界のプログラミングでは、注意⼒や専⾨知識
や知恵が必要となります。例えば、CAD(コンピュータによる設計⽀援)プログ
ラムの⼩さなバグが、⾶⾏機やダムの破滅的な崩壊を引き起こしたり、産業ロ
ボットの⾃⼰破壊を起こしたりするのです。
達⼈ソフトウェアエンジニアは、プログラムを構築する能⼒があり、結果

となるプロセスが意図するタスクを実⾏してくれることにある程度⾃信を持つ
ことができます。彼らは、システムの動作を前もって思い描くことができます。
プログラムを構造化して、予期しない問題が破滅的な結果を引き起こさないよ
うにする⽅法を知っています。そして、問題が起こってしまったときには、プ
ログラムをデバッグ (debug) できます。うまく設計されたコンピュータシステ
ムは、うまく設計された⾃動⾞や原⼦炉のように、モジュール化して設計され
ていて、部品が別々に作成、置き換え、デバッグできるようになっています。

Lispプログラミング
プロセスを記述するのには、適切な⾔語が必要です。この⽬的のために、私た
ちはプログラミング⾔語 Lispを使うことにします。私たちの⽇々の考えが普通
⾃然⾔語 (英語やフランス語や⽇本語など) によって表現されるように、そして
定量的な現象の記述が数学的記法によって表現されるように、私たちの⼿続き

的な思考は Lispによって表現されます。Lispは 1950年代の後半に、再帰⽅程
式 (recursion equation)と呼ばれるある種の論理表現についての推論を定式化
するものとして発明されたものです。この⾔語は John McCarthyにより考え
出され、彼の論⽂ “記号式の再帰⽅程式とそれらの機械による演算”(McCarthy
1960)に基づいています。
数学的な定式化としての⽣まれにもかかわらず、Lispは実⽤的なプログラ

ミング⾔語です。Lispインタプリタ (interpreter)は Lisp⾔語によって記述さ
れたプロセスを実⾏する機械です。最初の Lispインタプリタは、mit研究所の
⼈⼯知能部⾨の同僚、学⽣に⼿伝ってもらいながら、McCarthyが実装したも
のです。1 Lispという名前は LISt Processing(リスト処理)の頭⽂字で、この⾔
語は記号微分や代数式の積分のようなプログラミング上の問題に取り組むため
の記号操作能⼒を提供するために設計されたものです。この⽬的のために、ア
トムとリストとして知られる新しいデータオブジェクトを導⼊しています。こ
の特徴が、Lispがその時代のほかの⾔語から際⽴ったものとなっている最⼤の
特徴です。

Lispは、⼀致協⼒した設計の努⼒によってできたものではありません。ユ
ーザーの必要とすることや実⽤的な実装の検討事項に応えるものとして、実験
的なやり⽅で⾮公式に進化してきたものです。Lispの⾮公式な進化は何年も続
き、Lispユーザーのコミュニティは伝統的に、この⾔語の “公式な”定義を公布
しようとすることに抵抗してきました。この進化が⾔語の柔軟性と元々のエレ
ガントさと組み合わさって、広く使われている⾔語としては⼆番⽬に古い (こ
れより古いのは Fortranだけです)⾔語である Lispはずっと適応を続け、プロ
グラム設計についての最新の考え⽅を含むようになっています。そのため、今
では Lispは⽅⾔の集まりで、元々の特徴の多くを共通して持ちながらも、それ
ぞれ⼤きく違うこともあるものとなっています。この本で使う⽅⾔は Scheme
と呼ばれるものです。2

1Lisp 1 Programmer’s Manualは 1960年に登場し、Lisp 1.5 Programmer’s Manual
(McCarthy et al. 1965)は 1962年に出版されました。Lispの初期の歴史は、McCarthy
1978で説明されています。

21970年代に主要な Lispプログラムの⼤部分に⽤いられた 2つの⽅⾔は、mitのプ
ロジェクト macで開発された MacLisp (Moon 1978; Pitman 1983)と、Bolt Beranek
and Newman Inc.と Xerox Palo Alto研究センターで開発された Interlisp (Teitelman
1974)でした。Portable Standard Lisp (Hearn 1969; Griss 1981)、異なるマシンの間で
簡単に移植できるように設計された Lisp⽅⾔です。MacLispは、カリフォルニア⼤学
バークレー校で開発された Franz Lispや、mit⼈⼯知能研究所が Lispをとても効率よ
く実⾏できるよう設計した特定⽬的プロセッサ (処理器)に基づく Zetalisp (Moon and
Weinreb 1981)といった、いくつかの下位⽅⾔を⽣みました。この本で使⽤する Lisp⽅

実験的な性質を持ち、また記号操作に重点を置いているため、初期の Lisp
は数値計算については、⾮常に⾮効率的なものでした。少なくとも、Fortran
と⽐べるとそうでした。しかし、年を経るにつれ、プログラムを機械語に変換
して、⼗分効率的に数値計算を実⾏できるようにする Lispコンパイラが開発
されるようになりました。それに、特別な応⽤に対しては、Lispは⾮常に有効
に使われています。3 今でも Lispは、絶望的に⾮効率的だという昔からの評判
を乗り越えられていませんが、今では Lispは効率が最優先事項ではない数多
くの応⽤に使われています。例えば、Lispは OSのシェル⾔語や、エディタや
CADシステムの拡張⾔語として選ばれるものになっています。

Lispがメインストリームの⾔語でないとしたら、なぜそれをプログラミン
グの考察のための枠組みとして使うのでしょうか。それは、この⾔語が独特な
特徴を持っているため、プログラミングの重要な構成とデータ構造について学
び、それらを⽀える⾔語的特性と結びつけるのにとても便利な媒体だからです。
これらの特性の中でも最も意義深いものは、プロセスの Lispによる記述 (これ
は⼿続き (procedure)と呼ばれます)が、それ⾃⾝ Lispのデータとして表され、
操作できるということです。これが重要なのは、“受動的な”データと “能動的
な”プロセスという伝統的な区別を曖昧にする能⼒を使った強⼒なプログラム
設計のテクニックが存在するからです。これから⾒ていくように、Lispは⼿
続きをデータとして扱う柔軟性のおかげで、これらのテクニックを探求するの
に最も便利な⾔語のひとつになっています。⼿続きをデータとして表現する能
⼒は、コンピュータ⾔語を⽀えるインタプリタやコンパイラのような、ほかの
プログラムをデータとして操作しなければならないプログラムを書くのにも、
Lispをとても優れたものにしています。それに、こういったことを考えに⼊れ
なくても、Lispでのプログラミングは本当に楽しいのです。

⾔は Sheme (Steele and Sussman 1975)と呼ばれ、1975年に mit⼈⼯知能研究所のGuy
Lewis Steele Jr.とGerald Jay Sussmanにより開発され、のちに mitで教育⽬的で再実
装されたものです。Common Lisp⽅⾔ (Steele 1982, Steele 1990)は、初期の Lisp⽅⾔
の機能を組み合わせ、Lispの業界標準を作るために Lispコミュニティにより開発されま
した。Common Lispは 1994年に ansi標準 (ANSI 1994)になりました。

3そのような特別な応⽤のひとつは、⾃然科学的に重要な計算のブレークスルー— 太
陽系の運動についての積分の精度を⼆桁近く上げ、太陽系の⼒学はカオス的であること
を⽰した—でした。この計算は、新しい積分アルゴリズム、専⽤コンパイラ、そして専⽤
コンピュータによるものですが、これらはすべて Lispによって書かれたソフトウェアツ
ールの助けによって実装されたものです。(Abelson et al. 1992; Sussman and Wisdom
1992)

1.1 プログラミングの要素
強⼒なプログラミング⾔語は、コンピュータにタスクの実⾏を指⽰する道具と
いうだけではありません。⾔語は、私たちがプロセスについて考えをまとめる
枠組みとしても役に⽴ちます。そのため、⾔語を記述する際には、簡単な考え
を組み合わせてより複雑な考えを作るためにその⾔語が提供する⼿段に対して
特別な注意を払う必要があります。強⼒な⾔語は、これを達成するために、3
つのメカニズムを持っています。

• 基本式は、⾔語に関わる最も単純な実体を表します。
• 組み合わせ⽅法は、複合要素をより単純なものから構築する⽅法です。
• 抽象化⽅法は、複合要素に名前をつけ、単体として扱うための⽅法です。

プログラミングでは、⼆つの要素を扱います。それは、⼿続きとデータです (の
ちに、それらがそんなにはっきり分かれるものでないということがわかってき
ます)。⼤雑把に⾔うと、データは “物”で、私たちが操作したいものです。⼿
続きは、データを操作するための規則を記述したものです。そのため、強⼒な
プログラミング⾔語はすべて、基本的なデータや基本的な⼿続きを記述でき、
また⼿続きとデータを組み合わせ、抽象化する⽅法を備えていなければなりま
せん。
この章では、⼿続きを構築する規則に焦点を当てるため、単純な数値デー

タのみを扱うことにします。4 後の章では、複合データを扱う⼿続きを構築す
るのにも、これらの同じ規則が使えるということを⾒ていきます。

4数値を “単純なデータ”と⾒なすのは⽩々しいごまかしです。実際のところは、数値
の扱いは、どのプログラミング⾔語でも、⾮常に落とし⽳が多く混乱を招く⼀⾯となっ
ています。いくつかの典型的な問題は次のようなものです。コンピュータシステムによ
っては、2のような整数 (integer)と 2.71のような 実数 (real number)を別物として扱
います。実数 2.00は整数 2とは違うのでしょうか。整数に使われる算術演算は、実数に
使われるものと同じものでしょうか。6を 2で割ると、答えは 3でしょうか、それとも
3.0でしょうか。どれだけ⼤きな数値まで表現できるのでしょうか。⼩数点以下は何桁ま
で表すのでしょうか。整数の範囲は実数と同じでしょうか。これらの問題の先には、も
ちろん、丸め誤差や切り捨て誤差に関する問題の⼭が横たわっています。それは、数値
解析の科学そのものです。この本での焦点は⼤規模プログラム設計であり、数値計算の
テクニックではないので、これらの問題は無視することにします。この章での数値演算
の例では、⾮整数の演算では有限の⼩数点以下の精度を保つ算術演算を使い、そのよう
な場合に使われる通常の丸めの挙動が反映されています。

1.1.1 式
プログラミングを始める簡単な⽅法のひとつとして、Lispの Scheme⽅⾔のイ
ンタプリタと対話をしてみて、その結果を調べるというやり⽅があります。コ
ンピュータターミナルの前に座っているところを想像してみてください。あな
たが式 (expression)を⼊⼒すると、インタプリタはその式の評価 (evaluation)
の結果を表⽰することによって応答します。
タイプできる基本的な式のひとつとして、数値があります (より正確に⾔

うと、あなたがタイプする式は、10進数の数値を表す数値からなります)。Lisp
に次の数値を⼊⼒すると、
486

インタプリタは以下の内容を表⽰することで応答します。5

486

数値を表す式は、基本的な⼿続きを表す式 (例えば + や *)と組み合わせること
で複合式を作り、数値に対し⼿続きを適⽤することを表します。例えば:

(+ 137 349)
486

(- 1000 334)
666

(* 5 99)
495

(/ 10 5)
2

(+ 2.7 10)
12.7

このような式は、括弧の中の式のリストを区切ることにより作られ、⼿続きの
適⽤を意味するもので、組み合わせ (combination)と呼ばれます。リストの左

5この本全体を通して、ユーザーの⼊⼒とインタプリタが表⽰した応答との違いを強
調したい場合、後者を斜体で表します。

端の要素は 演算⼦ (operator)と呼ばれ、ほかの要素は被演算⼦ (operand)と呼
ばれます。組み合わせの値は、演算⼦によって指定された⼿続きを、被演算⼦
の値である 引数 (argument)に適⽤することによって得られます。
演算⼦を被演算⼦の左に置くというやり⽅は、前置記法 (prefix notation)

として知られているものですが、数学で慣例となっているやり⽅とは⼤幅に違
うので、最初は混乱するかもしれません。しかし、前置記法にはいくつかの利
点があります。その中のひとつは、任意の数の引数を取る⼿続きにも対応でき
るということです。例を以下に⽰します。
(+ 21 35 12 7)
75

(* 25 4 12)
1200

演算⼦はいつでも左端の要素で、組み合わせ全体は括弧で区切られているので、
曖昧さの⼊り込む余地はありません。
前置記法の⼆つ⽬の利点は、これを単純に拡張して、組み合わせをネスト

できる、つまり、組み合わせの要素がそれ⾃⾝組み合わせであるようなものが
作れるということです。
(+ (* 3 5) (- 10 6))
19

このようなネストや、Lispインタプリタが評価できる式の全体としての複雑性
には (原則的には)制限がありません。次のような、まだ⽐較的単純な式に出会
った場合、混乱してしまうのは私たち⼈間のほうです。
(+ (* 3 (+ (* 2 4) (+ 3 5))) (+ (- 10 7) 6))

インタプリタは、迷うことなく 57だと評価するでしょう。このような式は、次
のような形で書くことで、私たち⾃⾝にわかりやすいようにできます。
(+ (* 3

(+ (* 2 4)
(+ 3 5)))

(+ (- 10 7)
6))

プリティプリント (pretty-printing)として知られるフォーマットの慣習は、被
演算⼦が垂直に揃うようにそれぞれの⻑い組み合わせを書くというものなので

すが、上の式はそれに従って書いたものです。結果として、字下げが明確に式
の構造を⽰すことになります。6

複雑な式でも、インタプリタはいつでも同じ基本的なサイクルで動作しま
す。式を端末から読み、式を評価し、結果を表⽰するというものです。この操
作モードはよく、インタプリタのREPL(read-eval-print loop)モードと呼ばれ
ます。特に、結果を表⽰することをインタプリタに明⽰的に指⽰しなくてもい
いというところに注意してください。7

1.1.2 命名と環境
プログラミング⾔語の重要な特徴のひとつは、コンピュータ上のオブジェクト
を指すために名前を利⽤する⼿段を提供してくれるということです。このこと
を、値 (value)がそのオブジェクトである 変数 (variable)を名前によって特定
すると⾔います。

Lisp の Scheme ⽅⾔では、define(定義) によって対象に名前をつけます。
以下のように⼊⼒すると、
(define size 2)

インタプリタは sizeという名前と 2という値を関連づけます。8 sizeという
名前が⼀度 2という数値に関連づけられると、2という値を名前によって参照
できます。
size
2

(* 5 size)
10

6Lispシステムは、普通はユーザーが式を整形しやすいようにする機構を持っていま
す。その中でも特に役に⽴つものが⼆つあります。ひとつは、改⾏のタイミングで⾃動
的に正しいプリティプリントの位置にインデントするというもので、もうひとつは右括
弧が⼊⼒されるたびにマッチする左括弧をハイライトするというものです。

7Lispは、すべての式は値を持つという慣習に従っています。この慣習と、Lispは⾮
効率的な⾔語であるという昔からの評判を組み合わせて、Alan PerlisはOscar Wildeの
セリフをもじった次のような⽪⾁を⾔っています。“Lispプログラマはすべての値を知っ
ているが、コストについては何も知らない”

8この本では、定義を評価した後にインタプリタが返す応答を⽰すことはしません。そ
れは実装に強く依存するからです。

defineの使い⽅をさらに⾒ていきましょう。
(define pi 3.14159)
(define radius 10)
(* pi (* radius radius))
314.159
(define circumference (* 2 pi radius))
circumference
62.8318

defineは、この⾔語の持つ抽象化⽅法のうち、最も単純なものです。これによ
って、上で計算した circumference(円周)のような複合演算の結果を簡単な名
前で参照できます。⼀般的に、コンピュータ上のオブジェクトというものはと
ても複雑な構造を持っているので、使うたびにその細かいところを思い出して
打ち直さないといけないとしたら、とんでもなく⾯倒なことになります。イン
タプリタでは、この名前とオブジェクトの関連づけを⼀連の対話を通して少し
ずつ作っていくことができるので、このような段階的なプログラムの構築に特
に便利です。この特徴は、プログラムの開発・テストを少しずつ進めていくの
に向いていて、Lispプログラムが⼀般的に⼤量の⽐較的単純な⼿続きから構成
されるということの⼤きな理由となっています。
当然のことですが、インタプリタが値と記号を関連づけ、後から取り出すこ

とができるということは、名前とオブジェクトのペアを記録しておくために何
らかのメモリを持っておかないといけません。このメモリは環境 (environment)
と呼ばれます (より正確には、グローバル環境 (global environment)です。後で
学ぶように、演算には複数の異なる環境が使われることもあるからです)。9

1.1.3 組み合わせの評価
この章の⽬標のひとつは、⼿続き的に考えるうえでの問題点を分解することで
す。とりあえず、組み合わせを評価するにあたって、インタプリタはそれ⾃⾝、
⼿続きに従っているということを考えてみましょう。

組み合わせを評価するため、以下のことを⾏います。
1. 組み合わせの部分式を評価する

9第 3章では、インタプリタがどのように動作しているかを理解するため、またインタ
プリタを実装するために、この環境という概念がキーポイントになることを⽰します。

2. 部分式の左端 (演算⼦)の値となっている⼿続きを、引数 (被
演算⼦)、つまり部分式の残り値に適⽤する

こんな単純な規則からも、⼿続き⼀般についてのいくつかの重要なポイントが
わかります。⼀つ⽬のステップは、組み合わせに対する⼿続きを評価するには、
組み合わせのそれぞれの要素に対する評価⼿続きを先にやらないといけないと
いうことを⽰しています。そのため、評価規則は本質的に再帰的 (recursive)な
ものになります。これは、ステップのひとつとして、その規則⾃⾝を呼び出さ
ないといけないということです。10

再帰という考えが、深くネストした複合式をどれだけ簡潔に表現できるか
というところに注⽬してください。再帰がなければ、かなり複雑な⼿続きにな
るところです。例えば、次の式の評価する場合について考えます。
(* (+ 2 (* 4 6))

(+ 3 5 7))

この式を評価するには、4つの異なる組み合わせに対して評価規則を適⽤する
必要があります。この⼿続きは、図 1.1のように組み合わせを⽊の形で表すこ
とによってイメージできます。それぞれの組み合わせはノードとして表され、
そこから組み合わせの演算⼦と被演算⼦に対応する枝が⽣えています。終端ノ
ード (そこから⽣えている枝のないノード)は、演算⼦か数字を表しています。
評価を⽊という形で⾒ることで、被演算⼦の値が終端ノードから始まってそれ
ぞれのレベルで組み合わさりながら上に向かって伝わっていく様⼦がイメージ
できます。⼀般的に、階層的な⽊のようなオブジェクトを扱うためには、再帰
はとても強⼒なテクニックです。この “値を上に向かって伝える”という形の
評価規則は、⽊の集積 (tree accumulation)として知られています。
次に、⼀つ⽬のステップを繰り返して適⽤することで、ある点で評価する

対象が、組み合わせではなく数値や組み込み演算⼦やその他の名前といった基
本式になるというところに注⽬してください。このような基本的な場合につい
ては、以下のように規定することによって扱います。

• 数字の値は、それが⽰す値である
• 組み込み演算⼦の値は機械語の列で、それに対応する操作を実⾏する

10評価規則によると、⼀つ⽬のステップの⼀部として、組み合わせの左端の要素を評価
しなければいけないことになります。このことを奇妙に思われるかもしれません。この
時点では、左端の要素というと、+や *といった、⾜し算やかけ算のような組み込みの基
本的な⼿続きを表す演算⼦でしかありえないからです。のちに、演算⼦がそれ⾃⾝複合
式であるような組み合わせを使うことが便利だということを学びます。

図 1.1: 部分組み合わせの値を⽰した⽊表現

• その他の名前の値は、現在の環境でその名前に関連づけられたオブジェ
クトである

⼆つ⽬の規則は、三つ⽬の規則の特殊なケースと考えることができます。この
ことは、+と *といった記号もグローバル環境に含まれていて、⼀連の機械語
命令がそれらの “値”として関係づけられていると規定することによって可能
になります。ここでキーになるのは、式に出てくる記号の意味を決めるうえで
環境の果たす役割です。Lispのような対話的な⾔語では、(+ x 1)という式の
値について考えるのは、xという記号に意味を付与する環境についての情報が
なければ意味がありません。+という記号に対しても、意味を付与するのは環
境です。第 3章で学ぶように、評価を⾏う⽂脈を提供する存在としての環境と
いう⼀般的概念は、プログラムの実⾏について理解するうえで重要な役割を果
たしています。
上に書いた評価規則は、定義については扱っていないということに注意し

てください。例えば、(define x 3)を評価する際には、記号 xの値と 3という
⼆つの引数に defineを適⽤するということはしません。defineの⽬的は、ま
さに記号 xと値を関連づけるということだからです (つまり、(define x 3)は
複合式ではないということになります)。
このような、⼀般的評価規則に対する例外は、特殊形式 (special form)と呼

ばれます。ここまでの範囲では、defineは特殊形式の唯⼀の例ですが、ほかの
ものももうすぐ⾒ていくことになります。それぞれの特殊形式は、固有の評価
規則を持っています。それらのいろいろな種類の式 (それぞれ関連づけられた
評価規則を持つ)は、プログラミング⾔語の構⽂を構成します。ほかの多くの
プログラミング⾔語と⽐べると、Lispはとても単純な構⽂を持っています。そ
れは、式の評価規則はひとつの単純な⼀般規則と少数の特殊な形に対する特殊

規則からなるということです。11

1.1.4 複合⼿続き
ここまでで、強⼒なプログラミング⾔語であれば必ず持っているようないくつ
かの要素について、それが Lispにもあることを⾒てきました。

• 数値は基本データで、算術演算は基本⼿続きである。
• 組み合わせをネストすることで、演算を組み合わせることができる。
• 定義は名前と値を関連づけ、抽象化のためにある程度役に⽴つ。

ここでは、⼿続きの定義 (procedure definition)について学びます。これははる
かに強⼒な抽象化のテクニックで、複合演算に名前をつけ、それにひとつの単
位として参照できるようにするというものです。
まずは、“⼆乗の計算”とはどのように表現できるかを考えてみましょう。

例えば、“何かを⼆乗するには、その何かにその何か⾃⾝をかける”のようにな
るでしょう。これは、私たちの⾔語では次のように表すことができます。
(define (square x) (* x x))

これは、次のように理解できます。
(define (square x) (* x x))

| | | | | |
定義 ⼆乗する xを かける xを xで.

ここでは、複合⼿続き (compound procedure)を作り、それに squareという名
前をつけています。この⼿続きは、何かにそれ⾃⾝をかけるという演算を表し

11統⼀された書き⽅でも書けるものに対して、便利さのために別の表⾯構造を持た
せるような特殊な構⽂形式は、Peter Landin の造語を使って シンタックスシュガー
(syntactic sugar)と呼ばれることがあります。ほかの⾔語の使⽤者と⽐べると、Lispプ
ログラマは⼀般に、構⽂の問題をあまり気にしていません (これと対照的なのが Pascal
です。何でもいいので Pascalのマニュアルを開いて、どれだけのページが構⽂の記述に
充てられているか⾒てみてください)。このように構⽂を軽視しているのは、ひとつは表
⾯的な構⽂を変えやすい Lispの柔軟性のためです。もうひとつは、多くの “便利な”構
⽂構造が⾔語の統⼀性を損ない、プログラムが⼤きく複雑になるにつれ利点よりも⽋点
のほうが多くなるという観察結果のためです。Alan Perilの⾔葉を借りると、“シンタッ
クスシュガーはセミコロン (コロン (直腸)とかけている)の癌を引き起こす”ということ
です。

ています。かける数には xという名前をつけていますが、これは⾃然⾔語で代
名詞が果たすのと同じ役割を果たしています。この定義を評価すると、この複
合⼿続きを作成し、それを squareという名前と関連づけています。12

⼿続き定義の⼀般形式は以下の通りです。
(define (⟨名前⟩ ⟨仮引数⟩)
⟨本体⟩)

⟨名前⟩は、環境の中で⼿続きに関連づける記号です。13 ⟨仮引数⟩は、⼿続きの
本体の中で対応する引数を参照するために使う名前です。⟨本体⟩は、その中に
出てくる仮引数をその⼿続きが適⽤される実際の引数で置き換えた場合に、⼿
続き適⽤後の値を返すような式です。14 ⟨名前⟩と ⟨仮引数⟩は、定義する⼿続
きを実際に呼び出すときと同じように、括弧でくくります。

squareを定義したので、もうそれを使うことができます。
(square 21)
441
(square (+ 2 5))
49
(square (square 3))
81

squareは、ほかの⼿続きを定義するための構成部品として使うこともできま
す。例えば、x2 + y2 は次のように表現できます。
(+ (square x) (square y))

⼆つの数値が引数として与えられたときにその⼆乗の和を求める sum-of-
squares という⼿続きも、簡単に定義できます。

12ここで、⼆つの異なる操作が組み合わされていることに気をつけてください。まず
⼿続きを作成して、それに squareという名前をつけています。⼿続きを名前をつけずに
作るということと、すでに作られている⼿続きに名前をつけることということの、⼆つ
の概念を区別することは可能であり、重要なことでもあります。そのやり⽅については
1.3.2節で⾒ていきます。

13この本全体を通して、式の⼀般的な構⽂について記述する際には、⼭括弧でくくった
イタリックの記号 —例えば、⟨name⟩—を使って、それらの式を実際に使うときに埋めな
ければならない “スロット”を表します。

14もう少し⼀般化な⾔い⽅をすると、⼿続きの本体は式の列にもなりえます。その場
合、インタプリタは列のそれぞれの式を順番に評価し、最後の式の値を⼿続き適⽤の値
として返します。

(define (sum-of-squares x y)
(+ (square x) (square y)))

(sum-of-squares 3 4)
25

これで、sum-of-squaresをさらに別の⼿続きの構成部品として使うこともで
きるようになります。
(define (f a)

(sum-of-squares (+ a 1) (* a 2)))
(f 5)
136

複合⼿続きは、基本⼿続きとまったく同じように使うことができます。実際、
上に書いた sum-of-squaresの定義を⾒ても、squareが +や *のような組み込
み⼿続きなのか、複合⼿続きとして定義されたものなのか、⾒分けることはで
きないはずです。

1.1.5 ⼿続き適⽤の置換モデル
演算⼦が複合⼿続きを指すような組み合わせをインタプリタが評価するとき、
1.1.3節で説明したような、演算⼦が基本⼿続きを指す組み合わせを評価する
場合とほぼ同じような⼿順をたどります。つまり、インタプリタは組み合わせ
の各要素を評価し、⼿続き (組み合わせの演算⼦の値)を引数 (組み合わせの被
演算⼦の値)に適⽤するということです。
基本⼿続きを引数に適⽤する仕組みは、インタプリタに組み込まれている

と考えることができます。複合⼿続きについては、その適⽤⼿順は次のように
なります。

複合⼿続きを引数に適⽤するには、⼿続きの本体に出てくる仮引
数を対応する引数で置き換えて、それを評価する。

この⼿順の例として、次の組み合わせを評価してみましょう。
(f 5)

ここで、fは1.1.4節で定義した⼿続きです。まず、fの本体を取得することか
ら始めます。
(sum-of-squares (+ a 1) (* a 2))

次に、仮引数である aを、引数 5で置き換えます。
(sum-of-squares (+ 5 1) (* 5 2))

これによって、問題は⼆つの被演算⼦と sum-of-squaresという演算⼦の組み
合わせの評価ということになります。この組み合わせの評価は、三つの部分問
題を持っています。適⽤する⼿続きを得るためには演算⼦を評価する必要があ
り、引数を得るためには⼆つの被演算⼦を評価しなければなりません。ここで、
(+ 5 1) の結果は 6 で、(* 5 2) の結果は 10 なので、sum-of-squares とい
う⼿続きを 6と 10に適⽤することになります。これらの値によって sum-of-
squaresの本体に出てくる仮引数 xと yを置き換えて、次の式を得ます。
(+ (square 6) (square 10))

squareの定義を使うと、これは次の式になります。
(+ (* 6 6) (* 10 10))

かけ算によって、次のようになります。
(+ 36 100)

そして、最終的には次のようになります。
136

ここまでで説明した⼿順は、⼿続き適⽤の置換モデル (substitution model) と
呼ばれます。これは、この章に出てくる⼿続きに限ると、⼿続き適⽤の “意味”
を決めるモデルとして捉えることができます。しかし、⼆つ強調しておくこと
があります。

• 置換の⽬的は、⼿続き適⽤について考えやすくするためのもので、イン
タプリタが実際にどのように動いているかについて記述したものではあ
りません。インタプリタは普通、⼿続き適⽤の評価にあたって、⼿続きの
テキストを操作して仮引数を値で置き換えるということはしません。実
際には、この “置換”は仮引数のために局所環境を使うことによって実現
します。このことについては、第 3章と第 4章で、インタプリタの実装
について詳しく調べながら⾒ていきます。

• この本を通して、インタプリタの動作について、だんだん精巧になって
いく⼀連のモデルを提⽰していきます。最終的には、第 5章でインタプ
リタとコンパイラの完全な実装にまで到達します。置換モデルは、これ
らのモデルの最初のひとつ—評価⼿順について形式的に考えるための最

初の⼀歩となるもの— に過ぎません。⼀般に、科学や⼯学で現象をモデ
ル化するときには、単純化した不完全なモデルから始めます。物事を詳
細に調べていくにつれ、これらの単純なモデルは不適切になり、より精
密なモデルで置き換えなければならなくなります。置換モデルも例外で
はありません。特に、第 3章で “可変データ”を持つ⼿続きについて考え
る場合に、置換モデルが破綻してもっと複雑な⼿続き適⽤モデルによっ
て置き換えなければならないことがわかります。15

適⽤順序と正規順序
1.1.3節での評価の記述によると、インタプリタはまず演算⼦と被演算⼦を評
価し、評価結果として返される⼿続きを評価結果として返される引数に適⽤し
ます。これは、評価を⾏うための唯⼀のやり⽅ではありません。考えられるほ
かの評価モデルとしては、値が必要になるまで被演算⼦を評価しないというも
のがあります。その代わりに、まず被演算⼦の式を基本演算⼦しか出てこない
式になるまで置き換えてから評価を⾏います。この⽅法を使うと、(f 5)の評
価は次のような展開の連続によって進みます。
(sum-of-squares (+ 5 1) (* 5 2))
(+ (square (+ 5 1)) (square (* 5 2)) )
(+ (* (+ 5 1) (+ 5 1)) (* (* 5 2) (* 5 2)))

それから、次のように簡約されます。
(+ (* 6 6) (* 10 10))
(+ 36 100)

136

この⽅法でも前の評価モデルと同じ値が得られますが、⼿順が違います。特に、
こちらでは (+ 5 1)と (* 5 2)の評価がそれぞれ⼆回⾏われます。これは、式
(* x x)の xを、それぞれ (+ 5 1)と (* 5 2)で置き換えることに対応してい
ます。

15置換という考え⽅は単純なものですが、置換⼿順の厳密な数学的定義をしようとす
ると驚くほど複雑になるということがわかっています。この問題は、⼿続きの仮引数に
使われる名前と、⼿続きを適⽤する式で使われている (同じである可能性のある)名前と
を混同する可能性から来ています。実際に、論理学とプログラミング意味論の⽂献にお
いては、置換 (substitution)の間違った定義に関する⻑い歴史があります。置換に関する
精密な考察については、Stoy 1977を参照してください。

この “完全に展開してから簡約する”というもうひとつの評価⽅法は、正規
順序評価 (normal-order evaluation) として知られています。それに対して、イ
ンタプリタが実際に使っている “引数を評価してから適⽤する”という⽅法は、
適⽤順序評価 (applicative-order evaluation)と呼ばれます。置換によってモデ
ル化でき (この本の最初の⼆章の⼿続きはすべてそうです)、かつ正当な値を返
す⼿続き適⽤については、正規順序評価と適⽤順序評価は同じ値になるという
ことが証明できます (正規順序評価と適⽤順序評価が同じ値にならない “不当
な”値の例については、練習問題 1.5を参照してください)。

Lispは適⽤順序評価を使っています。理由のひとつは、上で⾒た (+ 5 1)
と (* 5 2) のような複数回の評価を避けることによる性能の向上です。そし
て、もっと⼤きな理由としては、置換によってモデル化できる⼿続きの範囲を
超えると正規順序評価がとても複雑になるということがあります。⼀⽅で、正
規順序評価は⾮常に価値のあるツールにもなりえます。そのことについては、
第 3章 と 第 4章で⼀部⾒ていきます。16

1.1.6 条件式と述語
この時点で私たちが定義できる種類の⼿続きは、表現⼒がとても乏しいもので
す。というのは、検査を⾏う⽅法がなく、また検査の結果によって異なる演算
を⾏うということもできないからです。例えば、次のような規則によって数値
が正か負かゼロかを検査して絶対値を求めるという⼿続きを定義することはで
きません。

|x| =


x if x > 0,

0 if x = 0,

−x if x < 0.

この構造は場合分け (case analysis)と呼ばれるもので、Lispにはそのような場
合分けを記述するための特殊な形式があります。それは cond (“conditional”(条
件的)の略)と呼ばれ、次のように使います。
(define (abs x)

(cond ((> x 0) x)
((= x 0) 0)

16第 3章ではストリーム処理 (stream processing)を導⼊します。これは、限定された
形の正規順序評価を組み⼊れることで、“無限”に⾒えるデータ構造を扱うという⼿法で
す。4.2節では、Schemeインタプリタに⼿を加え、Schemeの正規順序バージョンを作成
します。

((< x 0) (- x))))

条件式の⼀般形式は以下の通りです。
(cond (⟨p1 ⟩ ⟨e1 ⟩)

(⟨p2 ⟩ ⟨e2 ⟩)
. . .

(⟨pn ⟩ ⟨en ⟩))

これは、記号 condに続く、節 (clause)と呼ばれる括弧でくくった式のペアの
列によって構成されます。節の形式は次のようになります。
(⟨p⟩ ⟨e⟩)

それぞれのペアの最初の式は、述語 (predicate)—値が真か偽のどちらかとして
解釈される式—です。17

条件式は、次のように評価されます。まず、述語 ⟨p1 ⟩が評価されます。もし
その値が偽なら、⟨p2 ⟩が評価されます。もし ⟨p2 ⟩の値も偽なら、次は ⟨p3 ⟩が
評価されます。この⼿順は、値が真である述語が⾒つかるまで続きます。⾒つ
かると、インタプリタはその節に対応する結果式 (consequent expression) ⟨e⟩
の値を返します。⟨p⟩がどれも真でない場合、condの値は未定義となります。
述語 (predicate)という単語は、真か偽のどちらかに評価される式だけでな

く、真か偽のどちらかを返す⼿続きについても使われます。絶対値の⼿続き
absは、>, <, =という基本述語を使っています。18 これらは、⼆つの数値を引
数として取り、最初の数が⼆番⽬の数に⽐べて⼤きいか、⼩さいか、同じかと
いう検査を⾏い、その結果に従って真か偽かを返すというものです。
絶対値の⼿続きは、次のように書くこともできます。

(define (abs x)
(cond ((< x 0) (- x))

(else x)))

17“真か偽のどちらかとして解釈される”とは、以下のことを意味します。Schemeで
は、⼆つの特殊な値が存在し、#t、#fという定数として表されます。インタプリタが述
語の値をチェックするとき、#fを偽として解釈します。ほかの値はすべて真として扱わ
れます (つまり、論理的に⾔うと #tを定義する必要はないところですが、あったほうが
便利です)。この本では、#tと #fにそれぞれ関連づけた名前として、trueと falseを使
います。

18absは、このほかに “マイナス”演算⼦-も使っています。これは、(- x)のようにひ
とつの被演算⼦とともに使われる場合、符号の反転を表します。

これは、⽇本語で⾔うと “もし xがゼロより⼩さければ −xを返す。そうでな
ければ xを返す”となります。elseは特殊な記号で、condの最後の節の ⟨p⟩ の
代わりに使うことができます。これを使うと、それまでのすべての節がスキッ
プされたときに、これに対応する ⟨e⟩を condの値として返すようにできます。
実は、値が常に真として評価される式であれば何でも、ここの ⟨p⟩として使う
ことができます。
さらに、絶対値の⼿続きは次のように書くこともできます。

(define (abs x)
(if (< x 0)

(- x)
x))

ここでは、ifという特殊形式を使っています。これは、場合分けがちょうど⼆
つの場合からなる場合に使える、制約つきの条件式です。if式の⼀般形式は以
下の通りです。
(if ⟨predicate⟩ ⟨consequent⟩ ⟨alternative⟩)

if 式を評価するにあたって、インタプリタは最初に式の ⟨predicate⟩ の部分
を評価します。もし ⟨predicate⟩の評価結果が真である場合、インタプリタは
⟨consequent⟩を評価し、その値を返します。そうでなければ ⟨alternative⟩を評
価し、その値を返します。19

<や =, >のような基本述語に加えて、複合述語を構築するための論理複合
演算というものがあります。次の三つは、最もよく使われるものです。

• (and ⟨e1⟩ . . . ⟨en⟩)
インタプリタは式 ⟨e⟩を左から右にひとつずつ評価します。⟨e⟩のどれか
が偽と評価されると、and式の値は偽となり、残りの ⟨e⟩は評価されませ
ん。すべての ⟨e⟩が真と評価されると、and式の値は最後の式の値になり
ます。

• (or ⟨e1⟩ . . . ⟨en⟩)
インタプリタは式 ⟨e⟩を左から右にひとつずつ評価します。⟨e⟩のどれか
が真と評価されると、その値が or式の値として返され、残りの ⟨e⟩は評

19if と condのちょっとした違いとして condの各節の ⟨e⟩は式の列であってもいいと
いうことがあります。対応する ⟨p⟩が真になる場合、⟨e⟩内の式は順番に評価され、列
の最後の式の値が condの値として返されます。しかし、if式の中では ⟨consequent⟩と
⟨alternative⟩は単⼀の式でなければなりません。

価されません。すべての ⟨e⟩が偽と評価されると、or式の値は偽となり
ます。

• (not ⟨e⟩)
not式の値は、式 ⟨e⟩が偽と評価される場合は真になり、そうでなければ
偽となります。

andと orが特殊形式で、⼿続きではないということに注意してください。こ
れは、部分式がすべて評価されるとは限らないからです。notは通常の⼿続き
です。
これらの使い⽅の例として、数値 xが 5 < x < 10という値域にあるという

条件は、次のように表現できます。
(and (> x 5) (< x 10))

もうひとつの例として、ある数値がもうひとつの数値と⽐べて等しいかより⼤
きいかということを検査する述語は、以下のように定義できます。
(define (>= x y) (or (> x y) (= x y)))

または、次のように書くこともできます。
(define (>= x y) (not (< x y)))

練習問題 1.1: 以下の⼀連の式について、それぞれの式に対するイ
ンタプリタの応答として表⽰される結果はどうなるか。式の列は、
下記に⽰した順に評価されるとする。
10
(+ 5 3 4)
(- 9 1)
(/ 6 2)
(+ (* 2 4) (- 4 6))
(define a 3)
(define b (+ a 1))
(+ a b (* a b))
(= a b)
(if (and (> b a) (< b (* a b)))

b
a)

(cond ((= a 4) 6)
((= b 4) (+ 6 7 a))
(else 25))

(+ 2 (if (> b a) b a))

(* (cond ((> a b) a)
((< a b) b)
(else -1))

(+ a 1))

練習問題 1.2: 以下の式を前置記法に書き換えよ。
5 + 4 + (2− (3− (6 + 4

5
)))

3(6− 2)(2− 7)
.

練習問題 1.3: 三つの数値を引数として取り、そのうち⼤きいほう
から⼆つの数値の⼆乗の和を返す⼿続きを定義せよ。
練習問題 1.4: 我々の評価モデルでは、演算⼦が複合式であるよう
な組み合わせが作れるということを観察せよ。この観察結果を使
って、次の⼿続きのふるまいを説明せよ。
(define (a-plus-abs-b a b)

((if (> b 0) + -) a b))

練習問題 1.5: Ben Bitdiddleは、⾃分が対⾯しているインタプリ
タが適⽤順序評価と正規順序評価のどちらを使っているかを調べ
るテストを発明した。彼は、以下の 2つの⼿続きを定義する。
(define (p) (p))
(define (test x y)

(if (= x 0) 0 y))

それから、以下の式を評価する。
(test 0 (p))

適⽤順序評価を使うインタプリタでは、Benはどのようなふるま
いを観察するだろうか。正規順序評価を使うインタプリタでは、ど
のようなふるまいを観察するだろうか。あなたの回答を説明せよ
(特殊形式である ifの評価規則は、インタプリタが適⽤順序評価
でも正規順序評価でも同じであると仮定せよ。述語式が最初に評
価され、その結果によって結果式と代替式のどちらを評価するか
が決まる)。

1.1.7 例: ニュートン法による平⽅根
ここまでで⾒てきた通り、⼿続きは普通の数学の関数にとても似ています。ど
ちらも、ひとつ以上の引数によって決定される値を規定します。しかし、数学
の関数とコンピュータの⼿続きの間には、ひとつの重要な違いがあります。そ
れは、⼿続きは実効的なものでなければならないということです。
⼀例として、平⽅根を計算問題について考えてみましょう。平⽅根の関数

は、以下のように定義できます。
√
x = y ≥ 0 かつ y2 = x であるような y.

これは、完全に正当な数学関数です。これを使って、ある数値がほかの数値の
平⽅根かどうか判断することもできますし、平⽅根というもの全般についての
事実を導出することもできます。⼀⽅で、この定義は⼿続きを記述するもので
はありません。ある数値の平⽅根を実際にどうやって求めるかということにつ
いては、ほとんど何も教えてくれません。この定義を Lispっぽく書き換えても
どうにもなりません。
(define (sqrt x)

(the y (and (>= y 0)
(= (square y) x))))

これでは、ただの問題の先送りです。
関数と⼿続きの違いは、物事の属性について説明することと、どうやって

物事を⾏うかについて説明することの⼀般的な違いを反映したものです。これ
は、宣⾔的知識と命令的知識の違いと⾔われることもあります。数学では普通、
宣⾔的な (何であるか)記述が関⼼の対象ですが、コンピュータサイエンスでは
普通、命令的な (どうやるか)記述が関⼼の対象です。20

20宣⾔的記述と命令的記述は、数学とコンピュータサイエンスのように密接に関係し

平⽅根は、どうやって計算するのでしょうか。最も⼀般的な⽅法は、ニュ
ートンの逐次近似法を使うというものです。それによると、数値 xの平⽅根の
推定値として yがあるとき、yと x/yの平均を取るという単純な操作を⾏うこ
とでよりよい推定値 (実際の平⽅根により近い値)が得られます。21 例えば、2
の平⽅根は以下のように計算できます。推定値の初期値を 1とします。

推定値 商 平均
1 (2/1) = 2 ((2 + 1)/2) = 1.5
1.5 (2/1.5) = 1.3333 ((1.3333 + 1.5)/2) = 1.4167
1.4167 (2/1.4167) = 1.4118 ((1.4167 + 1.4118)/2) = 1.4142
1.4142 ... ...

この⼿順を続けると、平⽅根の近似解をどんどんよくしていくことができます。
さて、この⼿順を⼿続きという形で定式化してみましょう。まず、被開平

数 (平⽅根を計算したい数値)と推定値から始めます。推定値が⽬的に⼗分な程
度によければ、そこで終わりです。そうでなければ、改善した推定値を使って
⼿順を繰り返す必要があります。この基本戦略を、次のような⼿続きとして書
きます。
(define (sqrt-iter guess x)

(if (good-enough? guess x)
guess
(sqrt-iter (improve guess x) x)))

ています。例えば、プログラムによって計算された答えが “正しい”かどうかと⾔うこと
は、そのプログラムについての宣⾔的な⽂を作るということになります。これまでに、プ
ログラムの正しさを証明する技術を確⽴することを⽬的とした膨⼤な量の研究がありま
すが、このテーマの技術的な難しさは、命令的な⽂ (それによってプログラムを構築す
る)と宣⾔的な⽂ (物事を推理する)との間の移⾏について、うまく折り合いをつけると
いうことに関するものです。関連する話として、プログラミング⾔語設計の現在の重要
な領域のひとつは、いわゆる超⾼⽔準⾔語についての調査です。そのような⾔語では、宣
⾔的な⽂によって実際にプログラミングを⾏います。その発想となるものは、インタプ
リタを⼗分に⾼性能なものにして、“何であるか”という知識がプログラマによって与え
られると、“どうやるか”についての知識を⾃動的に⽣成できるようにしようというもの
です。これは⼀般的には不可能ですが、進歩のある重要な領域もあります。この発想に
ついては、第 4章で再度触れることにします。

21この平⽅根アルゴリズムは、実際にはニュートン法の特殊なケースです。ニュートン
法は、⽅程式の根を求めるための⼀般的な技法です。平⽅根アルゴリズム⾃体は、アレ
キサンドリアのヘロンにより a.d.1世紀に開発されたものです。⼀般的なニュートン法を
どのように Lispの⼿続きとして表すかについては、1.3.4節で⾒ていきます。

推定値を改善するには、古い推定値と、被開平数を古い推定値で割ったものと
の平均を取ります。
(define (improve guess x)

(average guess (/ x guess)))

averageは以下のように定義します。
(define (average x y)

(/ (+ x y) 2))

ここで、“⼗分によい”ということの定義も決める必要があります。次の⽅法は
説明のためには⼗分ですが、本当はあまりよいテストではありません。(練習問
題 1.7を参照してください)その発想は、答えの⼆乗と被開平数の差が事前に決
めておいた許容誤差 (ここでは 0.001とします)よりも⼩さくなるぐらいに答え
が近くなるまで、答えの改善を続けるというものです。22

(define (good-enough? guess x)
(< (abs (- (square guess) x)) 0.001))

最後に、始めるための⽅法が必要です。例えば、任意の数値の平⽅根について、
1という推定値を与えることもできます。23

(define (sqrt x)
(sqrt-iter 1.0 x))

これらの定義をインタプリタに⼊⼒すると、sqrtをほかの⼿続きと同じよう
に使えるようになります。

22述語には、述語であるということがわかりやすいように、疑問符で終わる名前をつ
けることが普通です。これはスタイル上の慣例にすぎません。インタプリタにとっては、
疑問符はただの通常の⽂字です。

23推定値の初期値を 1ではなく 1.0と表現していることに注意してください。多くの
Lispの実装では、この⼆つの間に何の違いもありません。しかし、mit Schemeは整数と
⼩数を区別し、整数同⼠の割り算は⼩数ではなく有理数を返します。例えば、10を 6で
割ると 5/3になりますが、10.0を 6.0で割ると 1.6666666666666667になります (有理数
の演算をどのように実装するかについては、2.1.1節で学びます)。もしこの平⽅根プログ
ラムの初期推定値を 1として、また xが整数である場合、平⽅根の計算でその後⽣成さ
れる値はすべて、⼩数ではなく有理数になります。有理数と⼩数の混合演算は常に⼩数
になるので、初期推定値を 1.0として始めると、その後計算される値はすべて強制的に⼩
数になります。

(sqrt 9)
3.00009155413138

(sqrt (+ 100 37))
11.704699917758145

(sqrt (+ (sqrt 2) (sqrt 3)))
1.7739279023207892

(square (sqrt 1000))
1000.000369924366

sqrtプログラムは、ここまでで紹介してきた単純な⼿続き型⾔語が、例えば C
や Pascalで書けるようなものであれば、どんな純粋な数値計算のプログラムで
も書けるということを⽰しています。コンピュータに何かを繰り返し実⾏させ
る繰り返し (ループ)という構造をこの⾔語に組み込んでいないことを考える
と、これは驚くべきことのように思えるかもしれません。しかし、sqrt-iter
は、繰り返しを達成することは⼿続きを呼ぶという通常の能⼒だけでできると
いうことの実例となっています。24

練習問題 1.6: Alyssa P. Hackerは、なぜ ifが特殊形式として提
供される必要があるのか理解できなかった。“condを使って普通
の⼿続きとして定義したらいいんじゃないの”と彼⼥は質問した。
Alyssaの友達である Eva Lu Atorは、これは実際にできると主張
し、ifの新しいバージョンを定義した。
(define (new-if predicate then-clause else-clause)

(cond (predicate then-clause)
(else else-clause)))

Evaは Alyssaにプログラムのデモを⾏った。
(new-if (= 2 3) 0 5)
5
(new-if (= 1 1) 0 5)
0

24繰り返しを実装するために⼿続き呼び出しを使うことの効率の問題について⼼配さ
れている読者の⽅は、1.2.1節の “末尾再帰”についての説明を⾒てください。

喜びながら、Alyssaは new-ifを使って平⽅根のプログラムを書き
直した。
(define (sqrt-iter guess x)

(new-if (good-enough? guess x)
guess
(sqrt-iter (improve guess x) x)))

Alyssaが平⽅根の計算するのにこれを使おうとすると、何が起こ
るだろうか。説明せよ。
練習問題 1.7: 平⽅根の計算に使った good-enough?テストは、と
ても⼩さい数値の平⽅根を求めるのにはあまり効率的ではないだ
ろう。また、実際のコンピュータでは、数値演算はほとんど必ず有
限の有効数字で⾏われることになる。この問題があるので、とて
も⼤きい数に対しては、我々のテストは不適切なものになる。こ
れらの⾔明について、⼩さい数と⼤きい数でテストがうまくいか
ない例を使って説明せよ。good-enough? を実装するためのもうひ
とつの戦略としては、guessが繰り返しごとにどれだけ変化する
かを確認して、差分が推定値に対してとても⼩さいものになった
ら⽌めるというものもある。これは、⼩さい数値や⼤きい数値に
対して、よりうまくいくだろうか。
練習問題 1.8: ⽴⽅根に対するニュートン法は、yが xの⽴⽅根の
近似値である場合、次の値によってよりよい近似を求めることが
できるという事実に基づいている。

x/y2 + 2y

3
.

この式を使って、平⽅根と同じような仕組みの⽴⽅根⼿続きを実
装せよ (1.3.4節では、平⽅根と⽴⽅根を抽象化したものとして、⼀
般的なニュートン法をどのように実装するかを⾒ていく)。

1.1.8 ブラックボックス抽象化としての⼿続き
sqrtは、相互に定義された⼿続きのセットとして定義されるプロセスの最初
の例です。sqrt-iterの定義が再帰的 (recursive)であることに注意してくださ
い。再帰的というのは、⼿続きがそれ⾃⾝を使って定義されているということ

図 1.2: sqrtプログラムの⼿続き分解

です。⼿続きを定義するのにその⼿続き⾃⾝を使えるという考えには不安にな
るかもしれません。そんな “循環的な”定義に何の意味があるのか、そんなもの
でコンピュータが実⾏できるような明確に定義されたプロセスを記述できるの
かというところがはっきりしないように思えるかもしれません。このことにつ
いては、1.2節でもっと詳しく⾒ていくことにします。ここでは、sqrtの例か
らわかる、ほかのいくつかの重要な点について考えてみましょう。
平⽅根を計算するという問題が、⾃然にいくつかの部分問題に分解されて

いるということに注意してください。推定値が⼗分によいかどうやって判断す
るか、どうやって推定値を改善するか、などです。これらのタスクは別々の⼿
続きによって成し遂げられます。sqrtプログラム全体は、図 1.2に⽰すような
⼿続きの塊として⾒ることができます。この構造は、問題を部分問題に分解す
る様⼦を反映しています。
この分解戦略の重要な点は、プログラムを部品に分けるというだけのこと

ではありません。分けるだけなら、どんな⼤きなプログラムでも分解はできま
す—最初の 10⾏、次の 10⾏、その次の 10⾏、といったやり⽅で。それより
重要なのは、それぞれの⼿続きが特定のタスクを成し遂げていて、それがほか
の⼿続きを定義する際にもモジュールとして使えるようになっているというこ
となのです。例えば、good-enough?⼿続きを squareを使って定義するとき、
square⼿続きは “ブラックボックス”と⾒なせます。その時点では、その⼿続
きがどうやって結果を計算するのかには関⼼がありません。関⼼があるのは、
それが⼆乗を計算するという事実だけです。⼆乗がどのように計算されるかと
いう詳細は隠しておいて、後で考えることにできます。実際、good-enough?⼿
続きに関しては、squareは⼿続きというより⼿続きを抽象化したもの、いわゆ
る⼿続き抽象 (procedural abstraction)と考えることができます。この抽象レベ
ルでは、⼆乗を計算するものであればどんな⼿続きでも同じことです。

そのため、返り値だけを考えると、⼆乗を計算する以下の⼆つの⼿続きは
区別不可能であるべきです。どちらもひとつの数値を引数として取り、その数
値の⼆乗を返り値として⽣成します。25

(define (square x) (* x x))
(define (square x) (exp (double (log x))))
(define (double x) (+ x x))

ですから、⼿続き定義は詳細を隠せるようになっていなければなりません。⼿
続きを使う⼈は⾃分でその⼿続きを書いたのではなく、別のプログラマからブ
ラックボックスとして取得したのかもしれないのです。⼿続きを使う⼈が、そ
れがどうやって実装されているかについての知識を求められるべきではありま
せん。

局所名
⼿続きを使う⼈が気にしないでいいようになっていなければならない実装上の
詳細としては、⼿続きを実装した⼈が仮引数の名前に何を選んだかということ
があります。そのため、次の⼆つの⼿続きは区別不可能であるべきです。
(define (square x) (* x x))
(define (square y) (* y y))

この原則—⼿続きの意味はその作者が使った仮引数の名前とは独⽴であるべき
ということ—は、ちょっと⾒ると⾃明なことのように思えますが、その影響範
囲は広汎にわたります。最も単純な影響としては、⼿続きの仮引数の名前は⼿
続き本体内に局所化されているべきだということです。例えば、平⽅根⼿続き
の中で、good-enough?の定義に squareを使いました。
(define (good-enough? guess x)

(< (abs (- (square guess) x))
0.001))

good-enough?の作者の意図は、第⼀引数の⼆乗が第⼆引数と⽐べて許容差の
範囲であるかを決定するということです。good-enough?の作者が、第⼀引数を

25どちらの⼿続きがより効率的な実装であるかということさえ、はっきりとはわかり
ません。それは利⽤できるハードウェアによります。機械によっては、“⾃明な”実装が
⾮効率的なものになることもあります。対数と逆対数の巨⼤なテーブルを⾮常に効率的
に格納している機械があると考えてみるのもよいでしょう。

参照するのに guessという名前を使い、第⼆引数を参照するのに x という名前
を使っていることがわかります。squareの引数は guessです。squareの作者
が (上の例のように)その引数を参照するのに xを使った場合、good-enough?
の xは squareの xとは別の xでなければならないということがわかります。
⼿続き squareの実⾏は、good-enough?で使われている xの値に影響を与えて
はいけません。その xの値は、squareが計算を終えた後にも good-enough?に
必要かもしれないからです。
仮引数がそれぞれの⼿続きの本体に局所化されていなかったとすると、

squareの仮引数 xは、good-enough?の仮引数 xと混同される可能性があるの
で、good-enough?のふるまいが、どのバージョンの squareを利⽤するかに依
存することになります。つまり、squareは私たちが望むようなブラックボック
スではなくなるでしょう。
⼿続きの仮引数は、その名前が何であってもかまわないという点で、⼿続

きの定義にとても特別な役割を果たしています。そのような名前は束縛変数
(bound variable)と呼ばれ、⼿続き定義は仮引数を束縛 (bind)するといった⾔
い⽅をします。束縛変数が⼿続き定義の中で⼀貫性を持って改名された場合、
⼿続きの意味は変わりません。26 もし変数が束縛されていなければ、それは⾃
由 (free)であると⾔います。束縛によって名前が定義される式のセットは、そ
の名前のスコープ (scope)と呼ばれます。⼿続き定義においては、その⼿続き
の仮引数として宣⾔された束縛変数は、その⼿続きの本体をスコープとして持
っています。
上記の good-enough?の定義の中で、guessと xは束縛変数ですが、<, -,

abs, squareは⾃由変数です。good-enough?の guessと xの代わりにどんな
名前を選んだとしても、それらが <, -, abs, squareと違うもので区別可能で
ある限り、good-enough?の意味とは独⽴しています (仮に、guessを absと改
名したとすると、変数 absをキャプチャ (capturing) するというバグを⼊れて
しまったことになります。そうすると、absが⾃由変数から束縛変数に変わっ
てしまいます)。しかし、good-enough?の意味は、それが使う⾃由変数の名前
と独⽴ではありません。その意味は、記号 absが数値の絶対値を求める⼿続き
を指しているという (この定義の外部の) 事実に実際に依存しています。good-
enough?の定義の中で、absを cosによって置き換えたとすると、違う関数を
計算することになります。

26⼀貫性のある改名という概念は、実際には扱いにくく、形式的な定義が難しいもので
す。有名な論理学者たちも、この点について恥ずかしい間違いを犯してきました。

内部定義とブロック構造
ここまでで、名前を分離する⽅法を⼀種類⼿に⼊れました。⼿続きの仮引数は
その⼿続きの本体に局所化されるというものです。平⽅根プログラムは、名前
の使い⽅をコントロールするやり⽅をもうひとつ⽰してくれます。現在のプロ
グラムは、次のような別々の⼿続きからなります。
(define (sqrt x)

(sqrt-iter 1.0 x))
(define (sqrt-iter guess x)

(if (good-enough? guess x)
guess
(sqrt-iter (improve guess x) x)))

(define (good-enough? guess x)
(< (abs (- (square guess) x)) 0.001))

(define (improve guess x)
(average guess (/ x guess)))

このプログラムの問題は、sqrtを使う⼈にとって重要な⼿続きは sqrtだけだ
ということです。ほかの⼿続き (sqrt-iter, good-enough?, improve)は余計な
もので、集中を妨げるだけです。また、平⽅根プログラムと組み合わせて動か
す別のプログラムの部品として good-enough?という⼿続きを定義したくても、
sqrtがその⼿続きを必要としているため、そうすることはできません。この問
題は、個別に働く多くのプログラマによって⼤きなシステムを構築する際に、
特に深刻なものになります。例えば、数値演算⼿続きの⼤きなライブラリを作
るときに、多くの数値関数は逐次近似法で計算され、そのために補助⼿続きと
して good-enough?や improveといった⼿続きを持っているかもしれません。
ここは、サブ⼿続きを局所化して sqrtの中に隠し、sqrtがほかの逐次近似法
の⼿続き (それぞれ⾃分のプライベートな good-enough?⼿続きを持っている)
と共存できるようにしたいところです。これを可能にするために、⼿続きが局
所的な内部定義を持てるようにします。例えば、平⽅根プログラムは次のよう
に書けます。
(define (sqrt x)

(define (good-enough? guess x)
(< (abs (- (square guess) x)) 0.001))

(define (improve guess x) (average guess (/ x guess)))
(define (sqrt-iter guess x)

(if (good-enough? guess x)
guess
(sqrt-iter (improve guess x) x)))

(sqrt-iter 1.0 x))

このような定義のネストはブロック構造 (block structure)と呼ばれます。これ
は、基本的には最も単純な名前パッケージング問題の正しい解決⽅法です。し
かし、ここにはもっといいアイデアが隠れています。補助⼿続きの定義を内部
化することに加えて、それらを単純にすることもできます。xは sqrtに束縛さ
れているので、sqrtの内部で定義された good-enough?, improve, sqrt-iter
は xのスコープ内にあります。つまり、xをこれらの⼿続きに明⽰的に渡す必
要はないということです。そうする代わりに、次に⽰すように内部定義では x
を⾃由変数にします。そうすると、xの値は、それを囲む⼿続きである sqrtの
引数から取ってくることができます。このような規定は レキシカルスコーピン
グ (lexical scoping)と呼ばれます。27

(define (sqrt x)
(define (good-enough? guess)

(< (abs (- (square guess) x)) 0.001))
(define (improve guess)

(average guess (/ x guess)))
(define (sqrt-iter guess)

(if (good-enough? guess)
guess
(sqrt-iter (improve guess))))

(sqrt-iter 1.0))

この本では、⼤きなプログラムを扱いやすい部品に分割するために、ブロッ
ク構造を多⽤します。28 ブロック構造という考え⽅は、プログラミング⾔語
Algol 60に端を発します。ブロック構造はほとんどの先進的なプログラミング
⾔語に存在し、⼤きなプログラムの構築を体系化するための重要なツールとな
っています。

27レキシカルスコーピングにおいては、⼿続きの⾃由変数はその⼿続きを包む⼿続き
定義の束縛を参照すると規定します。つまり、⼿続きが定義された環境での探索が⾏わ
れるということです。第 3章で、環境とインタプリタの詳細なふるまいについて学ぶと
きに、この仕組みがどう働くのか詳細に⾒ていくことにします。

28組み込み定義は、⼿続き本体の最初に来なければなりません。定義とその使⽤をごち
ゃごちゃに混ぜたようなプログラムの実⾏については責任を負いかねます。

1.2 ⼿続きとそれが⽣成するプロセス
ここまで、プログラミングのさまざまな要素について考えてきました。基本的
な算術演算を使い、その演算を組み合わせ、その組み合わせた演算を複合⼿続
きと定義して抽象化を⾏いました。しかし、それだけではプログラミングのや
り⽅を知っているとは⾔えません。私たちの状況は、チェスの駒がどう動くか
というルールを覚えたものの、定跡や戦術、戦略については何も知らない⼈と
同じです。チェスの初⼼者のように、そのドメイン内でのよくあるパターンに
ついてまだ知らないということです。私たちは、どの⼿を指すのがいい⼿なの
か (どの⼿続きを定義するのがいいのか)という知識を持っていません。指し⼿
(⼿続きの実⾏)の結果を予想する経験も⾜りません。
考慮中のアクションについて結果を思い描く能⼒は、達⼈プログラマにな

るために決定的に重要なことです。統合的・創造的な活動であれば、何でもそ
うです。例えば、熟練した写真家になるためには、シーンの⾒⽅を学び、露出
と現像の条件の選び⽅によってそれぞれ領域がどれだけの濃さで写真に表れる
かを学ばなければなりません。そうすることによって初めて、望む効果を得る
ためにはフレーム、光の当て⽅、現像などをどのように計画すればいいかを逆
算できるようになるのです。それはプログラミングでも同じことです。プログ
ラミングにおいては、プロセスが取るべき処理の流れを計画し、プログラムと
いう⼿段でプロセスをコントロールします。達⼈になるためには、さまざまな
タイプの⼿続きが⽣み出すプロセスを思い描けるようになる必要があります。
そのようなスキルを⾝につけて初めて、望むふるまいをするプログラムを確実
に構築できるようになるのです。
⼿続きは、計算プロセスの局所展開 (local evolution)のためのパターンで

す。それは、プロセスの各段階が、どのように前の段階の上に構築されるかを
記述するものです。プロセスの局所展開は⼿続きによって記述されるものです
が、できればプロセスの全体的なふるまい、別の⾔い⽅をするとグローバル
(global)なふるまいについても何か⾔えるようになりたいところです。これは、
⼀般的にはとても難しいことです。しかし、プロセスの展開のよくあるパター
ンについての説明を試みるぐらいはできるでしょう。
この節では、単純な⼿続きによって作られたプロセスの、いくつかの⼀般

的な “形”について⾒ていきます。また、これらのプロセスが時間や空間といっ
た重要な計算リソースをどういった⽐率で消費していくのかについても調べて
いきます。ここで考える⼿続きはとても単純なものです。これは、写真撮影で
テストパターンが果たす役割と同じ役割を果たしています。つまり、独⽴した
実⽤的な例というよりは、極度に単純化したプロトタイプパターンです。

(factorial 6)
(* 6 (factorial 5))
(* 6 (* 5 (factorial 4)))
(* 6 (* 5 (* 4 (factorial 3))))
(* 6 (* 5 (* 4 (* 3 (factorial 2)))))
(* 6 (* 5 (* 4 (* 3 (* 2 (factorial 1))))))
(* 6 (* 5 (* 4 (* 3 (* 2 1)))))
(* 6 (* 5 (* 4 (* 3 2))))
(* 6 (* 5 (* 4 6)))
(* 6 (* 5 24))
(* 6 120)
720 

図 1.3: 6!を求めるための線形再帰プロセス

1.2.1 線形再帰と反復
まず、階乗関数について考えてみましょう。定義は以下の通りです。

n! = n · (n− 1) · (n− 2) · · · 3 · 2 · 1.

階乗を計算する⽅法は数多くあります。ひとつの⽅法は、任意の正の整数 nに
対して、n!は nと (n − 1)!をかけたものと等しいということを利⽤したもの
です。

n! = n · [(n− 1) · (n− 2) · · · 3 · 2 · 1] = n · (n− 1)!.

そのため、(n− 1)!を計算し、その答えに nをかけることで n!を計算できると
いうことになります。これに 1!は 1に等しいという規定を加えると、そのまま
⼿続きに変換できます。
(define (factorial n)

(if (= n 1)
1
(* n (factorial (- n 1)))))

この⼿続きが 6!の計算を実⾏する様⼦を観察するのに、1.1.5節の置換モデル
を使うことができます。そうすると、図 1.3のようになります。

(factorial 6)
(fact-iter 1 1 6)
(fact-iter 1 2 6)
(fact-iter 2 3 6)
(fact-iter 6 4 6)
(fact-iter 24 5 6)
(fact-iter 120 6 6)
(fact-iter 720 7 6)
720

図 1.4: 6!を求めるための線形反復プロセス

ここで、階乗計算を別の⾓度から⾒てみましょう。n!の計算規則として、最
初に 1×2を⾏い、それから答えに 3をかけ、それから 4をかけ、nまで続ける
というものに決めることもできます。もう少し形式的に書くと、計算中の積と、
1から nまで数え上げていくカウンタとの両⽅を保存するということです。計
算については、次の規則に従ってカウンタと積がステップごとに同時に変化す
るというように記述できます。
product ← counter * product
counter ← counter + 1

そして、n!とはカウンタが nを超えた時点での積の値であると規定します。
ここでまた、問題の記述を階乗計算の⼿続きとして書き直します。29

(define (factorial n)
(fact-iter 1 1 n))

29実際のプログラムでは、前の節で紹介したブロック構造を使って fact-iterの定義
を隠すところです。
(define (factorial n)

(define (iter product counter)
(if (> counter n)

product
(iter (* counter product)

(+ counter 1))))
(iter 1 1))

ここでは、⼀度に考えることを最⼩限にするために、そうすることを避けています。

(define (fact-iter product counter max-count)
(if (> counter max-count)

product
(fact-iter (* counter product)

(+ counter 1)
max-count)))

前回と同じく、6! の計算プロセスを視覚化するために置換モデルを使い、図
1.4として⽰します。

2つのプロセスを⽐べてみましょう。ある観点からは、ほとんど違いがない
ように⾒えます。どちらも、同じ定義域の同じ数学関数を計算しています。n!
を求めるのに nに⽐例したステップ数が必要だという点も同じです。それどこ
ろか、どちらのプロセスも同じ順番のかけ算を⾏い、同じ順番で部分積を得ま
す。⼀⽅で、プロセスの “形”について考えると、これらがまったく違う展開の
仕⽅をしていることがわかります。
⼀つ⽬のプロセスについて考えてみましょう。置換モデルは、図 1.3の⽮印

で⽰されるような、展開に続く縮約という形を表しています。展開は、プロセス
が 遅延演算 (deferred operation)の連鎖 (ここではかけ算の連鎖)を構築する際
に起こります。縮約は、演算が実際に⾏われる際に起こります。遅延演算の連
鎖によって特徴づけられるこのタイプのプロセスは、再帰プロセス (recursive
process)と呼ばれます。このプロセスを実⾏するには、インタプリタは後で実
⾏する演算について記録しておく必要があります。n!の計算では、遅延乗算の
連鎖の⻑さ、またそれを記録するために必要な情報の量は、nに対して線形に
(nに⽐例して)増加します。ステップ数も、同じように線形に増加します。この
ようなプロセスは、線形再帰プロセス (linear recursive process)と呼ばれます。
それとは対照的に、⼆つ⽬のプロセスは伸びたり縮んだりしません。各ステッ

プで記録しておく必要があるのは、どんな nに対しても、変数 product, counter,
max-count の現在の値だけです。これは反復プロセス (iterative process)と呼び
ます。⼀般的に、反復プロセスとは状態が限られた数の状態変数 (state variable)
に集約されるようなプロセスです。それに、プロセスの状態が変化するととも
に状態変数がどのように更新されるかという固定した規則と、どのような条件
でプロセスを停⽌するかを定めた (必須ではない)終了テストが加わります。n!
の計算では、必要なステップ数は nに対して線形に増加します。このようなプ
ロセスは、線形反復プロセス (linear iterative process)と呼ばれます。
この⼆つのプロセスの違いは、別の⾒⽅もできます。反復プロセスの場合、

プログラムの変数はどの時点でもプロセスの状態を完全に記述しています。ス
テップの間で計算を⽌めたとして、計算を再開するのに必要なのは、三つのプ

ログラム変数の値をインタプリタに与えることだけです。再帰プロセスではそ
うはいきません。この場合、インタプリタに保持されていてプログラム変数に
は含まれていない、いくつかの “隠された”情報が追加で存在し、遅延演算の連
鎖を扱ううえで “プロセスがどの地点にいるか”を指し⽰しています。連鎖が
⻑くなると、保持するべき情報もそれにつれて⼤きくなります。30

反復と再帰を⽐較する際に、再帰プロセス (process)という概念と再帰⼿続
き (procedure)という概念を混同しないように気をつける必要があります。⼿
続きについて再帰と⾔うときは、その⼿続きの定義が (直接または間接的に)そ
の⼿続き⾃⾝を参照しているという構⽂的事実を指しています。しかし、プロ
セスが例えば線形再帰パターンになっていると⾔う場合は、どのようにプロセ
スが展開するかということであって、⼿続きがどのように書かれているかとい
う構⽂の問題ではありません。fact-iterのような再帰⼿続きについて、それ
が反復プロセスを⽣成すると⾔うのは、ちょっと気になるかもしれません。し
かし、このプロセスは実際に反復的です。状態は三つの状態変数によって完全
に表され、プロセスを実⾏するためにインタプリタは三つの変数を記録してお
くだけで⼗分です。
プロセスと⼿続きの区別が厄介なのは、⼀般的な⾔語 (Adaや Pascal、C⾔

語を含む)のほとんどの実装では、再帰⼿続きの解釈には必ず⼿続き呼び出し
の回数に⽐例してメモリを消費するように設計されているからです。その⼿続
きが原理的に反復的である場合であっても、そうなってしまいます。その結果、
そのような⾔語では、反復⼿続きを記述するには専⽤の “ループ構造”に頼ら
なければなりません。do, repeat, until, for, whileなどがその例です。第 5
章で考える Schemeの実装には、この⽋点がありません。反復プロセスは、再
帰⼿続きとして記述されていても、固定の空間で実⾏できます。この性質を持
った実装は 末尾再帰 (tail-recursive)と呼ばれます。末尾再帰の実装では、反復
は普通の⼿続き呼び出しメカニズムを使って表現できるので、特別な反復構造
は必要なく、シンタックスシュガーとして役に⽴つ程度です。31

30第 5章でレジスタマシン上での⼿続きの実装について考えるとき、どんな反復プロ
セスでも、固定のレジスタセットを持ち補助メモリを持たない機械として “ハードウェ
アで”実現できるということを⾒ていきます。それに対し、再帰プロセスを実現するに
はスタック (stack)として知られる補助的なデータ構造が必要です。

31末尾再帰は、コンパイラの最適化トリックとして⻑い間知られてきました。Carl
Hewitt (1977)は、末尾再帰に⼀貫した意味論上の基礎を与えました。その説明は、“メ
ッセージパッシング”という計算モデルによるもので、それについては 第 3章で考察し
ます。これに影響を受けて、Gerald Jay Sussmanと Guy Lewis Steele Jr. (Steele and
Sussman 1975参照)は Schemeのための末尾再帰インタプリタを構築しました。Steele
は後に、末尾再帰が⼿続き呼び出しを⾃然にコンパイルすることの結果であることを⽰

練習問題 1.9: 次の⼆つの⼿続きは、どちらも inc, decという⼿続
きによって⼆つの正の整数を加算する⽅法を定義している。⼿続
き incは引数を 1増やし、decは引数を 1減らす。
(define (+ a b)

(if (= a 0) b (inc (+ (dec a) b))))
(define (+ a b)

(if (= a 0) b (+ (dec a) (inc b))))

置換モデルを使って、それぞれの⼿続きが (+ 4 5)を評価する際
に⽣成するプロセスを図⽰せよ。これらのプロセスは反復だろう
か、それとも再帰だろうか。
練習問題 1.10: 次の⼿続きは、アッカーマン関数と呼ばれる数学
の関数を計算する。
(define (A x y)

(cond ((= y 0) 0)
((= x 0) (* 2 y))
((= y 1) 2)
(else (A (- x 1) (A x (- y 1))))))

以下の式の値は何になるか。
(A 1 10)
(A 2 4)
(A 3 3)

Aが上で定義された⼿続きであるとき、以下の⼿続きについて考
えよ。
(define (f n) (A 0 n))
(define (g n) (A 1 n))
(define (h n) (A 2 n))
(define (k n) (* 5 n n))

⼿続き f, g, hによって、正の整数 nに対して計算される関数に対
し、簡潔な数学的定義を与えよ。例えば、(k n)は 5n2を計算する。

しました (Steele 1977)。Schemeの ieee規格は、Schemeの実装が末尾再帰であること
を要求しています。

1.2.2 ⽊の再帰
計算するうえでよく使われるパターンとして、ここでは新たに ⽊の再帰 (tree
recursion)を紹介します。例として、フィボナッチ数列の計算について考えて
みましょう。この数列では、それぞれの数値はその直前にある⼆つの数値の合
計になっています。

0, 1, 1, 2, 3, 5, 8, 13, 21, . . . .

⼀般に、フィボナッチ数は次のような規則で定義できます。

Fib(n) =


0 n = 0 のとき
1 n = 1 のとき
Fib(n− 1) + Fib(n− 2) それ以外

この定義は、フィボナッチ数列を計算する再帰⼿続きに直接変換できます。
(define (fib n)

(cond ((= n 0) 0)
((= n 1) 1)
(else (+ (fib (- n 1))

(fib (- n 2))))))

この計算のパターンについて考えてみます。(fib 5)を計算するには、(fib 4)
と (fib 3)を計算します。(fib 4)を計算するには、(fib 3)と (fib 2)を計
算します。⼀般に、展開されたプロセスは、図 1.5で⽰すように⽊のような形を
しています。枝は、(底以外の)それぞれのレベルで⼆つに分かれています。こ
れは、fib⼿続きが呼ばれるたびに⾃⾝を⼆回呼び出すということを反映して
います。
この⼿続きは、⽊の再帰のプロトタイプになっていますので、学習のため

には役に⽴ちます。しかし、フィボナッチ数を計算する⽅法としてはひどいも
のです。余計な計算が多すぎるからです。図 1.5で、(fib 3)の計算が—それ
は全体の半分ぐらいの仕事ですが— 完全に重複してしまっていることがわか
るでしょうか。さらに⾔うと、この⼿続きが (fib 1)や (fib 0)を計算する回
数 (⼀般に、上記の⽊の中での葉の数)が正確に Fib(n + 1)になることは簡単
に⽰せます。このやり⽅がどれだけひどいかということは、Fib(n)の値が nに
対して指数的に増えるということを考えればわかるでしょう。より正確には、
Fib(n)は φn/

√
5に最も近い整数になります (練習問題 1.13参照)。φは次の値

                            fib 5

               fib 4                      fib 3

        fib 3          fib 2          fib 2       fib 1

                                                 1
   fib 2    fib 1   fib 1   fib 0   fib 1    fib 0

             1      1      0      1       0
fib 1   fib 0

 1      0

図 1.5: (fib 5)の計算時に⽣成される⽊の再帰プロセス

です。
φ =

1 +
√
5

2
≈ 1.6180

φは⻩⾦⽐ (golden ratio)で、次の等式を満たします。
φ2 = φ+ 1.

そのため、このプロセスのステップ数は、⼊⼒に対して指数的に増加しま
す。⼀⽅で、必要な空間は⼊⼒に対して線形にしか増加しません。計算のどの
時点でも、⽊の中で上にあるノードだけを記録しておけばいいからです。⼀般
に、⽊の再帰のプロセスで必要なステップ数は⽊のノード数に⽐例し、必要な
空間は⽊の最⼤の深さに⽐例します。
フィボナッチ数の計算は、反復プロセスとして定式化することもできます。

考え⽅は、aと bという整数のペアを使い、Fib(1) = 1, Fib(0) = 0と初期化し
て、次の変換を同時に適⽤することを繰り返すというものです。

a ← a+ b,

b ← a.

この変換を n回適⽤した後、aと bがそれぞれ Fib(n+1)と Fib(n)に等しくな

るということは簡単に⽰せます。つまり、以下の⼿続きによって、フィボナッ
チ数を反復的に計算できるということです。
(define (fib n)

(fib-iter 1 0 n))
(define (fib-iter a b count)

(if (= count 0)
b
(fib-iter (+ a b) a (- count 1))))

こちらの、Fib(n)を計算する⼆つ⽬の⽅法は線形反復です。⼆つの⽅法で
必要なステップ数の違いは—ひとつは nに対して線形、もうひとつは Fib(n)
そのものと同じ割合— ⼩さな⼊⼒でも、とても⼤きなものになります。
このことから、⽊の再帰のプロセスが役に⽴たないと結論づけてはいけま

せん。数値ではなく、階層的な構造を持つデータを操作するプロセスについて
考える場合、⽊の再帰は⾃然で強⼒なツールだということがわかります。32 し
かし、数値演算の場合であっても、⽊の再帰のプロセスはプログラムを理解し、
設計するのに役に⽴ちます。例えば、⼀つ⽬の fib⼿続きは⼆つ⽬よりずっと
⾮効率的ですが、ずっと素直な書き⽅で、ほとんどフィボナッチ数列の定義を
Lispに変換しただけのようなものです。反復アルゴリズムを定式化するのは、
計算が三つの状態変数を使った反復として書き直せることに気づく必要があり
ます。

例: 両替パターンの計算
反復的フィボナッチアルゴリズムを思いつくには、ほんのちょっとの⼯夫で⼗
分でした。それに対して、次の問題を考えてみてください。1ドルを両替する
やり⽅はいくつあるでしょうか。使うのは、50セント、25セント、10セント、
5セント、1セントのコインです。より⼀般的には、任意の⾦額に対して、両替
のパターン数を計算する⼿続きを書くことはできるでしょうか。
この問題には、再帰⼿続きによる簡単な解き⽅があります。使えるコイン

の種類が、何らかの決まった順番を持つとしましょう。すると、次の関係が成
り⽴ちます。

n種類のコインを使って⾦額 aを両替するやり⽅のパターン数は、以下の
合計となる。

321.1.3節は⽊の再帰の例となっています。インタプリタ⾃⾝、⽊の再帰のプロセスを
使って式を評価しているのです。

• ⼀つ⽬の種類のコイン以外のすべての種類のコインを使って⾦額 aを両
替するやり⽅のパターン数。

• n種類の硬貨すべてを使って、⾦額 a − dを両替するやり⽅のパターン
数。dは、⼀つ⽬の種類のコインの額⾯とする。

どうしてこれが⾔えるのかを考えるには、両替のやり⽅が⼆つのグループに分
けられることに注⽬します。⼀つ⽬の種類のコインを使わないものと、使うも
のです。そのため、ある⾦額に対する両替のパターン数は、⼀つ⽬の種類のコ
インを使わないでその⾦額を両替するやり⽅の数と、⼀つ⽬の種類のコインを
使って両替するやり⽅の数の合計になります。しかし、後者のパターン数は、
最初の種類のコインを⼀枚使った後の、残りの⾦額に対する両替のパターン数
と等しくなります。
このようにすると、ある⾦額に対する両替の問題を、より⼩さな⾦額をよ

り少ない種類のコインで両替する問題に再帰的に⼩さくしていくことができま
す。この⼩さくする規則を注意深く考えてみて、次の単純な場合について決め
ておけば、それがアルゴリズムの記述に使えるということを確認してください。
33

• もし aがちょうど 0なら、両替パターンは 1と数える。
• もし aが 0未満なら、両替パターンは 0と数える。
• もし nが 0なら、両替パターンは 0と数える。

この記述は、再帰⼿続きに簡単に翻訳できます。
(define (count-change amount) (cc amount 5))
(define (cc amount kinds-of-coins)

(cond ((= amount 0) 1)
((or (< amount 0) (= kinds-of-coins 0)) 0)
(else (+ (cc amount

(- kinds-of-coins 1))
(cc (- amount

(first-denomination
kinds-of-coins))

kinds-of-coins)))))
(define (first-denomination kinds-of-coins)

33例えば、10セントの両替を 5セントと 1セントを使って⾏う場合、⼩さくする規則
がどのように適⽤されるか、詳細にやってみてください。

(cond ((= kinds-of-coins 1) 1)
((= kinds-of-coins 2) 5)
((= kinds-of-coins 3) 10)
((= kinds-of-coins 4) 25)
((= kinds-of-coins 5) 50)))

(first-denomination⼿続きは、利⽤可能な硬貨の種類の数を⼊⼒として、⼀
つ⽬の種類のコインの額⾯を返します。ここでは、硬貨は額⾯の⼤きなものか
ら⼩さなものという順番で並んでいるとしていますが、どのような順番でも同
じようにうまくいきます) これで、1ドルの両替という元々の質問に答えるこ
とができます。
(count-change 100)
292

count-changeは、fibの⼀つ⽬の実装同様、冗⻑な⽊の再帰のプロセスを⽣成
します (292の計算にはかなりの時間がかかるでしょう)。⼀⽅で、答えを計算
するためのよりよいアルゴリズムをどう設計するかというのは⾃明ではありま
せん。この問題は、読者への宿題とします。⽊の再帰のプロセスはとても⾮効
率的になりうるものの、記述と理解が簡単であることが多いので、“賢いコンパ
イラ”を設計して、⽊の再帰のプロセスを同じ結果を計算するより効率的な⼿
続きに変換できるようにしたら両者のいいとこ取りができるんじゃないかとい
う提案がされています。34

練習問題 1.11: 関数 f は次のように定義される。

f(n) =

{
n if n < 3,
f(n− 1) + 2f(n− 2) + 3f(n− 3) if n ≥ 3.

f を演算する⼿続きを再帰プロセスによって書け。また、f を演算
する⼿続きを反復プロセスによって書け。

34冗⻑な計算に対処するひとつのアプローチは、値が計算されるときに⾃動的にテーブ
ルが構築されるようにするというものです。ある引数に⼿続きを適⽤するよう要求され
るたびに、まずその値がすでにテーブルに格納されているか確認します。格納されていれ
ば、冗⻑な計算を実⾏しないようにします。この戦略は、テーブル化 (tabulation)や メモ
化 (memoization)として知られるもので、素直に実装できます。テーブル化は、(count-
changeのような)指数的なステップ数の必要なプロセスを、必要な時間と空間が⼊⼒に
対して線形に増加するプロセスに変換するのにも時々使われます。練習問題 3.27参照。

練習問題 1.12: 以下の数値のパターンは パスカルの三⾓形 (Pas-
cal’s triangle)と呼ばれる。

1
1 1

1 2 1
1 3 3 1

1 4 6 4 1
. . .

三⾓形の辺の数値はすべて 1で、三⾓形内部の数値は、それぞれ
その上にある 2つの数値の合計となっている。35 パスカルの三⾓
形の要素を再帰プロセスによって求める⼿続きを書け。
練習問題 1.13: Fib(n)が φn/

√
5に最も近い整数であることを証

明せよ。φ = (1+
√
5)/2とする。ヒント：ψ = (1−

√
5)/2と置く。

帰納法とフィボナッチ数の定義 (1.2.2節参照)を使って、Fib(n) =
(φn − ψn)/

√
5であることを証明せよ。

1.2.3 増加オーダー
前節の例は、プロセスが計算リソースを消費する量の増え⽅に⼤幅に差が出る
ことがあるということを⽰しています。この違いを記述する便利な⽅法のひと
つとして、増加オーダー (order of growth)という概念によって、⼊⼒が増える
に従ってプロセスが必要とするリソースがどの程度になるかについての⼤雑把
な感覚をつかむということがあります。
問題の⼤きさを測るパラメータを nとして、⼤きさ nの問題についてプロ

セスが必要とするリソースの量を R(n)とします。前節の例では、ある関数が
計算を⾏う対象とする数値を n としましたが、ほかの可能性もあります。例え
ば、もし⽬的がある数値の平⽅根の近似値を求めることであれば、求める精度
の桁数を nとするのもいいでしょう。⾏列のかけ算であれば、n を⾏列の⾏数

35パスカルの三⾓形の要素は⼆項係数 (binomial coefficient)と呼ばれます。これは、n
番⽬の⾏が (x+ y)n を展開したときの各項の係数となっているからです。この係数計算
のパターンは、Blaise Pascalが 1653年に書いた確率理論の偉⼤な論⽂である Traité du
triangle arithmétiqueに書かれたものです。Knuth (1973)によると、同様のパターンは
1303年に中国の数学者朱世傑により出版された Szu-yuen Yü-chien (四元⽟鑑)や、12
世紀のペルシャの詩⼈であり数学者でもあった Omar Khayyamの著作や、12世紀イン
ドの数学者 Bháscara Ácháryaの著作にも現れるということです。

としてもいいかもしれません。⼀般に、問題というものは、プロセスについて
分析するために都合のいい属性をいくつか持っているものです。同じように、
R(n)は使⽤される内部ストレージのレジスタ数であったり、実⾏される基本
的な機械語命令数であったりします。⼀度に決まった数の命令しか実⾏しない
コンピュータでは、実⾏時間は実⾏する基本的な機械語命令数に⽐例します。
任意の⼗分に⼤きな n に対して、n と独⽴な正の定数 k1 と k2 が存在し、

k1f(n) ≤ R(n) ≤ k2f(n)を満たすとき、R(n)は増加オーダーが Θ(f(n))であ
ると⾔い、R(n) = Θ(f(n))(“シータ f(n)”と読む)と書きます。(別の⾔い⽅を
すると、⼤きな nに対して、R(n)の値は k1f(n)と k2f(n)で挟まれるという
ことです)
例えば、1.2.1節で記述した階乗を求める線形再帰プロセスでは、ステップ

数は⼊⼒ nに⽐例します。つまり、このプロセスに必要なステップ数はΘ(n)で
増加します。必要な空間もΘ(n)で増加するということを確認しています。反復
による階乗計算では、ステップ数は依然として Θ(n)ですが、空間は Θ(1)—つ
まり、定数です。36 ⽊の再帰によるフィボナッチ数の計算は、Θ(φn)ステップ
と空間 Θ(n)を必要とします。ここで、φは1.2.2節で⽰した⻩⾦⽐です。
増加オーダーは、プロセスのふるまいについて⼤雑把な記述をするだけで

す。例えば、n2 が必要なプロセス、1000n2が必要なプロセス、3n2 + 10n+ 17

ステップが必要なプロセスは、すべては Θ(n2)という増加オーダーになりま
す。⼀⽅で、問題の⼤きさを変えた場合のプロセスのふるまいをどう予測する
かということについては、増加オーダーは役に⽴つ指標となります。Θ(n)の
(線形の)プロセスはで、問題の⼤きさを 2倍にすると、使うリソースもだいた
い 2倍になります。指数的プロセスは、問題サイズが 1増えるたびに、リソー
ス使⽤量は定数倍になります。1.2節の残りで、増加オーダーが対数的であり、
問題の⼤きさが 2倍になると必要リソースが定数量だけ増える 2つのアルゴリ
ズムを⾒ていきます。

練習問題 1.14: 1.2.2節の count-change⼿続きによって 11セント
に対する両替のやり⽅を求める際に⽣成されるプロセスを図⽰す
る⽊を描け。両替する⾦額の増加に対して、このプロセスが使う
空間とステップ数はどのような増加オーダーになるか。

36この書き⽅の背後には、⾮常に多くの過度の単純化が隠れています。例えば、プロセ
スのステップ数を “機械語命令”で数えるとすると、例えばかけ算を実⾏するのに必要な
機械語命令の数はかける数値の⼤きさと独⽴であると想定しています。これは、数値が
⼗分に⼤きければ成り⽴ちません。空間の⾒積もりについても同じようなことが⾔えま
す。プロセスの設計や記述と同じように、プロセスの分析もさまざまな抽象化レベルで
⾏うことができます。

練習問題 1.15: (ラジアンで指定される) ⾓度の正弦は、xが⼗分
に⼩さいとき sinx ≈ x の近似式によって計算できる。また、次の
三⾓法の恒等式によって、sinの引数の⼤きさを⼩さくできる。

sinx = 3 sin x
3
− 4 sin3 x

3

(この練習問題では、⾓が “⼗分に⼩さい”とはその⼤きさが 0.1ラ
ジアン以下であることとする) これらの考えは、以下の⼿続きに組
み込まれている。
(define (cube x) (* x x x))
(define (p x) (- (* 3 x) (* 4 (cube x))))
(define (sine angle)

(if (not (> (abs angle) 0.1))
angle
(p (sine (/ angle 3.0)))))

a. (sine 12.15) を評価する際に、⼿続き p は何回適⽤され
るか。

b. (sine a)が評価される際に sine⼿続きによって⽣成される
プロセスが使⽤する空間とステップ数の増加オーダーを (a
の関数として)求めよ。

1.2.4 指数計算
与えられた数値の指数を求める問題について考えましょう。基数 bと正の整数
の指数 nを引数に取り、bnを求める⼿続きということになります。ひとつのや
り⽅として、これは次のように再帰的定義によって定義できます。

bn = b · bn−1,

b0 = 1,

これは、すぐに次の⼿続きに変換できます。
(define (expt b n)

(if (= n 0)
1
(* b (expt b (- n 1)))))

これは線形再帰プロセスで、Θ(n)のステップ数と Θ(n)の空間を必要としま
す。階乗と同じように、等価な線形反復としてすぐに定式化できます。
(define (expt b n)

(expt-iter b n 1))
(define (expt-iter b counter product)

(if (= counter 0)
product
(expt-iter b

(- counter 1)
(* b product))))

こちらのバージョンは、Θ(n)のステップ数と Θ(1)の空間を必要とします。
指数は、⼆乗を連続して使うことによって、より少ないステップ数で計算

できます。例えば、b8 を次のように計算するのではなく、

b · (b · (b · (b · (b · (b · (b · b)))))) ,
3回の乗算で求めることができます。

b2 = b · b,
b4 = b2 · b2,
b8 = b4 · b4.

この⽅法は、指数が 2の冪乗である場合にうまくいきます。また、次の規則を
使うことで、⼀般的な指数計算に⼆乗の連続を使うことができます。

bn = (bn/2)2 n が偶数の場合,
bn = b · bn−1 n が奇数の場合.

この⽅法は、⼿続きとして表現できます。
(define (fast-expt b n)

(cond ((= n 0) 1)
((even? n) (square (fast-expt b (/ n 2))))
(else (* b (fast-expt b (- n 1))))))

ここで、ある整数が偶数かどうかをテストする述語は、基本⼿続きの remainder
(割り算の余り)を使って次のように定義します。
(define (even? n)

(= (remainder n 2) 0))

fast-exptにより展開されるプロセスは、空間・ステップ数ともに、nに対し
て対数的に増加します。このことは、fast-exptを使って b2n を計算するのは
bnの計算よりかけ算を 1回多く実⾏するだけだということからわかります。こ
のため、かけ算の回数が 1回増えるたびに、計算できる指数の⼤きさは (ほぼ)
倍になります。したがって、n乗の計算をするのに必要なかけ算の回数は、だ
いたい 2を底とした nの対数と同じぐらいの速さで増加することになります。
このプロセスは Θ(logn)の増加オーダーを持っています。37

Θ(logn)での増加と Θ(n)での増加の違いは、nが⼤きくなるにつれて驚く
ようなものになります。例えば、fast-exptは n = 1000に対して 14回の乗算
しか必要としません。38 ⼆乗の連続という考え⽅を使って、対数的ステップ
数で指数を計算する反復アルゴリズムを作ることもできます (練習問題 1.16参
照)。しかし、反復アルゴリズムにはよくあることですが、再帰アルゴリズムほ
ど素直に書き下すことはできません。39

練習問題 1.16: fast-exptと同じように⼆乗の連続を使って、指
数計算を対数的ステップ数で実⾏する反復的プロセスを⽣成する
⼿続きを設計せよ。(ヒント：(bn/2)2 = (b2)n/2を使い、指数 n、基
数 bのほかに追加の状態変数 aを持ち、積 abn が各状態を通して
⼀定であるような状態変換を定義せよ。プロセスの最初では aは 1
とし、答えはプロセス終了時の aの値として得られる。⼀般的に、
各状態を通して⼀定であるような不変量 (invariant quantity)を定
義するというテクニックは、反復アルゴリズムを設計するうえで
強⼒な⽅法である。)

練習問題 1.17: この節の指数計算アルゴリズムは、指数をかけ算
の繰り返しによって計算することを基本としている。同じような

37より正確には、必要なかけ算の回数は、底を 2とした nの対数から 1を引いたもの
に nの⼆進数表現での 1の数を⾜したものになります。この合計は、底を 2とした nの
対数の 2倍より常に⼩さくなります。オーダー記法の定義には k1 と k2 という任意の定
数が出てきますが、このことは対数的なプロセスについて対数の底は問題にならないと
いうことを意味します。そのため、対数的プロセスはすべて Θ(logn)として記述されま
す。

381000乗の計算なんか誰もしないだろうとお考えかもしれません。1.2.6節を参照して
ください。

39この反復アルゴリズムは古代から存在します。紀元前 200 年より前に Áchárya
Pingala によって書かれた Chandah-sutraに記載されています。Knuth 1981の 4.6.3節
には、この⽅法を含むさまざまな指数関数の計算⽅法についての詳細な考察と分析があ
ります。

⽅法で、整数のかけ算を⾜し算の繰り返しによって実⾏すること
もできる。次のかけ算⼿続き (私たちの⾔語では⾜し算しかでき
ず、かけ算ができないと仮定して)は expt⼿続きと同じような仕
組みになっている。
(define (* a b)

(if (= b 0)
0
(+ a (* a (- b 1)))))

このアルゴリズムは、bに対して線形のステップ数を取る。ここで、
⾜し算のほかに、整数を倍にする doubleという演算と、(偶数の)
整数を 2で割る halveという演算があるとする。これらを使って、
fast-exptと同じような仕組みで対数的ステップ数を取るかけ算
⼿続きを設計せよ。
練習問題 1.18: 練習問題 1.16と 練習問題 1.17の結果を使って、⾜
し算、double、halveによって、かけ算を対数的ステップ数で実⾏
する反復的プロセスを⽣成する⼿続きを考えよ。40

練習問題 1.19: フィボナッチ数を対数的ステップ数で計算する巧
妙なアルゴリズムがある。1.2.2節の fib-iterプロセスでの状態
変数 aと bの変換、a← a+ b、b← aを思い出そう。この変換を T

と呼ぶことにする。すると、1と 0から始めて T を繰り返し n回
適⽤すると Fib(n + 1)と Fib(n)というペアができることになる。
⾔い換えれば、フィボナッチ数はペア (1, 0)に対して Tn(変換 T

の n乗)を適⽤することで求められるということだ。ここで、(a, b)
というペアを a ← bq + aq + ap と b ← bp + aq に変換するような
変換族 Tpq の特殊な場合 (p = 0、q = 1)として T を考えることに
する。このような変換 Tpq を⼆回適⽤すると、その効果は同じ形
式を持つ Tp′q′ を⼀回適⽤するのと同じであることを⽰し、pと q

に対する p′ と q′を求めよ。こうすると、この種の変換に対する⼆
乗を明⽰的に⾏うことができるようになり、fast-expt ⼿続きと
同じように、Tn を⼆乗の連続によって求められるようになる。こ

40このアルゴリズムは、“ロシア農⺠法”と呼ばれることもあるかけ算の⽅法で、古く
から存在します。その使⽤例は、最も古い⼆つの数学⽂献のひとつ、リンド数学パピル
スにも⾒られます。これは、紀元前 1700年ごろにエジプトの書記官 A’h-moseによって
書かれた (それも、より古い⽂献から写本された)ものです。

れらをすべて組み合わせて、対数的ステップ数で動く次の⼿続き
を完成させよ。41

(define (fib n)
(fib-iter 1 0 0 1 n))

(define (fib-iter a b p q count)
(cond ((= count 0) b)

((even? count)
(fib-iter a

b
⟨??⟩ ; p’ を計算する
⟨??⟩ ; q’ を計算する
(/ count 2)))

(else (fib-iter (+ (* b q) (* a q) (* a p))
(+ (* b p) (* a q))
p
q
(- count 1)))))

1.2.5 最⼤公約数
2つの整数 aと bの Greatest Common divisor(gcd:最⼤公約数)は aと bの両
⽅をぴったり割り切れる最⼤の整数と定義されます。例えば、16と 28 の gcd
は 4です。第 2章で、有理数演算をどのように実装⽅法するかを検討する際に、
有理数を既約にするために gcdの計算が必要になります (有理数を既約にする
ためには、分⺟と分⼦をそれらの gcdで割ります。例えば、16/28は 4/7にな
ります)。2つの整数の gcdを求める⽅法のひとつとしては、それらを素因数分
解して共通の素因数を探すというものがありますが、それよりずっと効率的な
有名なアルゴリズムが存在します。
そのアルゴリズムの考え⽅は、rが aを bで割った余りであれば、aと bの

公約数は bと rの公約数とまさに同じものになるという観察に基づいています。
このため、次の等式

GCD(a,b) = GCD(b,r)
41この練習問題は Joe Stoyによって提案されたもので、Kaldewaij 1990での例をもと
にしています。

によって、gcdを計算する問題を、より⼩さな整数ペアの gcdを計算する問題
へと連続的に簡約していくことができます。例えば、

GCD(206,40) = GCD(40,6)
= GCD(6,4)
= GCD(4,2)
= GCD(2,0)
= 2

この例では gcd(206, 40)を gcd(2, 0)まで簡約しています。その答えは 2です。
任意の 2つの正の整数から始めても、繰り返し簡約していくことで最終的には
必ず 2つ⽬の数値が 0であるペアにできるということが証明可能です。そのと
き、gcdはペアのもう 1つの値です。gcdを求めるこの⽅法は、ユークリッド
の互除法 (Euclid’s Algorithm)として知られています。42

ユークリッドの互除法は、⼿続きとして簡単に表現できます。
(define (gcd a b)

(if (= b 0)
a
(gcd b (remainder a b))))

これは反復プロセスを⽣成し、ステップ数は計算する数値に対して対数的に増
加します。
ユークリッドの互除法が必要とするステップ数が対数的に増加するという

事実は、フィボナッチ数と興味深い関係を持っています。
Lamé の定理: ユークリッドの互除法によって、ある数値ペアの
gcdを計算するのに k ステップを必要とする場合、ペアの⼩さい
ほうの数値は k番⽬のフィボナッチ数以上である。43

42ユークリッドの互除法は、ユークリッドの Elements(紀元前 300年ごろ、第 7巻)に
出てきたためにそう呼ばれます。Knuth (1973)によると、⾃明でないアルゴリズムとし
ては知られている限り最古のものと考えられるそうです。(練習問題 1.18)で出てきた古
代エジプトのかけ算⽅法は確かにこれよりも古いのですが、Knuthが説明するように、
ユークリッドのアルゴリズムは、具体例の集まりとしてではなく⼀般的なアルゴリズム
として⽰されたものとしては最古のものです。

43この定理は、1845年に Gabriel Laméによって証明されたものです。彼はフランス
の数学者かつエンジニアで、主に数理物理学への貢献で知られています。この定理を
証明するには、ユークリッドの互除法が k ステップで停⽌する、ak ≥ bk であるペア
(ak, bk)について考えます。証明は (ak+1, bk+1) → (ak, bk) → (ak−1, bk−1)が簡約プロ

この定理を使って、ユークリッドの互除法の増加オーダーを得ることができま
す。⼿続きの nを⼆つの⼊⼒のうちの⼩さいほうだとします。プロセスが kス
テップかかるとすると、n ≥ Fib(k) ≈ φk/

√
5が必ず成り⽴ちます。したがっ

て、ステップ数 kは nに対して (φを底とする)対数的に増加します。つまり、
増加オーダーは Θ(logn)となります。

練習問題 1.20:ある⼿続きが⽣成するプロセスは、もちろん、イ
ンタプリタの規則によって異なる。例として、上で説明した反復
gcd⼿続きについて考える。この⼿続きを、1.1.5節で考察した正
規順序評価によって解釈するとする (ifに対する正規順序評価規
則は練習問題 1.5に記述されている)。置換法を (正規順序に対し
て)使って、(gcd 206 40)を評価する際に⽣成されるプロセスを
図⽰し、実際に実⾏される remainder演算を⽰せ。(gcd 206 40)
の正規順序評価では、remainder演算は何回実⾏されるだろうか。
適⽤順序評価ではどうだろうか。説明せよ。

1.2.6 例: 素数判定
この節では、整数 nの素数性をチェックする⼆つの⽅法について記述します。
ひとつは増加オーダーが Θ(

√
n)で、もうひとつは増加オーダーが Θ(logn)の

“確率的”なアルゴリズムです。節末の練習問題では、これらのアルゴリズムに
基づいたプログラミングのプロジェクトを提案します。
セスの中の連続する 3つのペアである場合、必ず bk+1 ≥ bk + bk−1 であるという主張
に基づいています。この主張を証明するには、簡約ステップが ak−1 = bk, bk−1 =“ak
を bk で割った余り”という変換を適⽤することとして定義されていることについて考
えます。2 つ⽬の等式は、ak = qbk + bk−1 が成り⽴つある正の整数 q が存在するこ
とを意味しています。ここで、q は最⼩でも 1なので、ak = qbk + bk−1 ≥ bk + bk−1

が成り⽴ちます。しかし、⼀つ前の簡約ステップから、bk+1 = ak です。そのため、
bk+1 = ak ≥ bk + bk−1 が成り⽴ちます。これで主張は証明できました。ここで、アル
ゴリズムが停⽌するのに必要なステップ数である k に対して数学的帰納法を使うことで
定理が証明できます。k = 1のとき、これは bが Fib(1) = 1以上であればよいので、真
です。次に、k以下のすべての整数について定理が真であると仮定し、k + 1の場合につ
いて導きます。(ak+1, bk+1) → (ak, bk) → (ak−1, bk−1)が簡約プロセスにおける連続
するペアであるとします。数学的帰納法の仮定より、bk−1 ≥ Fib(k − 1)と bk ≥ Fib(k)
が成り⽴ちます。ここで、先ほど証明した主張とフィボナッチ数の定義を適⽤すると、
bk+1 ≥ bk + bk−1 ≥ Fib(k) + Fib(k − 1) = Fib(k + 1)が導けます。これで Laméの定
理の証明が完成します。

約数を探す
古代から、数学者は素数に関する問題に魅了されてきました。そして、多くの
⼈々が数値が素数であるかテストする⽅法に取り組んできました。ある数値が
素数であるかテストする⽅法のひとつとして、その数値の約数を探すというも
のがあります。次のプログラムは、ある数値 nをちょうど割り切る (1以上の)
最も⼩さな数を⾒つけます。このプログラムは、2から始まる⼀連の数字 nに
ついて割り切れるかどうかを調べるという素直な⽅法でそれを⾏っています。
(define (smallest-divisor n) (find-divisor n 2))
(define (find-divisor n test-divisor)

(cond ((> (square test-divisor) n) n)
((divides? test-divisor n) test-divisor)
(else (find-divisor n (+ test-divisor 1)))))

(define (divides? a b) (= (remainder b a) 0))

数値が素数であるかは次のようにテストできます: nは、n⾃⾝がその最⼩の約
数である場合、かつその場合に限り、素数である。
(define (prime? n)

(= n (smallest-divisor n)))

find-divisorの終了条件は、もし nが素数でないならば、それは √n以下の
約数を持つという事実に基づいています。44 つまり、このアルゴリズムは、1
から√nまでの約数についてだけテストすればよいということです。そのため、
nが素数であるかを判定するのに必要なステップ数は、増加オーダーが Θ(

√
n)

となります。

フェルマーテスト
Θ(logn)の素数判定は、フェルマーの⼩定理として知られる整数論の結果に基
づいています。45

44もし dが nの約数であれば、n/dも約数になります。しかし、dと n/dの両⽅とも√
nより⼤きいということはありえません。
45Pierre de Fermat (1601-1665)は、近代整数論の創始者と考えられています。彼は多
くの重要な整数論上の結果を発⾒しましたが、彼は普通その結果だけを公表し、証明を
与えませんでした。フェルマーの⼩定理は、彼が 1640年に書いた⼿紙に記録されてい
ます。最初の公開された証明は、オイラーによって 1736年に与えられました (また、そ
れより早い同じ証明がライプニッツの未発表の草稿に発⾒されています)。フェルマーの

フェルマーの⼩定理: nが素数で、aが nより⼩さい任意の正の整
数であるとき、aの n乗は法 nに関して aと合同である。

(⼆つの数値は、その両⽅を nで割ったときの余りが同じになる場合、法 nに
関して合同 (congruent modulo n)と⾔います。また、aを nで割ったときの余
りは法 nに関する a の剰余、または単純に a modulo nと呼ばれます。)
もし nが素数でなければ、⼀般に a < nであるほとんどの aは上記の関係

を満たしません。このことから、次の素数判定アルゴリズムが導かれます。あ
る数値 nが与えられたとき、a < n である適当な数値 a を取り、法 nに関す
る an の剰余を求めます。もし結果が aと等しくなければ、nは確実に素数で
はありません。もし aと等しければ、nは素数かもしれません。今度は、別の
適当な数値 aを取り、同じ⽅法でテストを⾏います。それもまた等式を満たせ
ば、nが素数であることについてより確信が持てるようになります。たくさん
の aについて試験を⾏っていくことで、結果についての確信を増やしていくこ
とができます。このアルゴリズムは、フェルマーテストとして知られています。
フェルマーテストを実装するには、ある数値の冪乗の、別のある数値を法

とした剰余を求める⼿続きが必要です。
(define (expmod base exp m)

(cond ((= exp 0) 1)
((even? exp)
(remainder
(square (expmod base (/ exp 2) m))
m))

(else
(remainder
(* base (expmod base (- exp 1) m))
m))))

これは、1.2.4節の fast-expt⼿続きにそっくりです。⼆乗の連続を使ってい
るため、ステップ数は指数に対して対数的に増加します。46

結果の中で最も有名なものは—フェルマーの最終定理として知られているものですが—
1637年に、彼の持っていた (3世紀のギリシャ⼈数学者 Diophantusによる)Arithmetic
という本に “私は真に驚くべき証明を発⾒したが、書き残すにはこの余⽩は狭すぎる”と
いう注記をつけてメモされていたものです。フェルマーの最終定理の証明を⾒つけるこ
とは、整数論の中で最も有名な課題のひとつとなりました。完全な解答は、1995年にプ
リンストン⼤学の Andrew Wilesによってついに与えられました。

46指数 e が 1 より⼤きい場合の簡約ステップは、任意の整数 x,y, m に対して、xy

フェルマーテストは、1以上 n− 1以下のランダムな数値 aを選び、aの n

乗の modulo nが aに等しいかをチェックすることで⾏います。ランダムな数
値 aは⼿続き randomを使って選びます。randomは Schemeの基本⼿続きに含
まれているとします。randomは、受け取った整数よりも⼩さい⾮負整数を返し
ます。そのため、1から n− 1の乱数を得るために randomに n− 1を渡し、返
り値に 1を⾜します。
(define (fermat-test n)

(define (try-it a)
(= (expmod a n n) a))

(try-it (+ 1 (random (- n 1)))))

次の⼿続きは、引数で与えられた数値の回数だけテストを実⾏します。テスト
がすべて成功すれば trueを、そうでなければ falseを返します。
(define (fast-prime? n times)

(cond ((= times 0) true)
((fermat-test n) (fast-prime? n (- times 1)))
(else false)))

確率的⼿法
ほとんどの馴染みのあるアルゴリズムは正しいことが保証された答えを返すも
のですが、フェルマーテストはその点で性質が異なります。フェルマーテスト
では、得られた答えは答えの正しさは確率的なものでしかありません。より正
確には、nがフェルマーテストに失敗すれば、n が素数でないことは確実です。
しかし、nがテストをパスするということは、⾮常に強い指標ではありますが、
それでも nが素数であるという保証にはなりません。ここで主張したいと思っ
ていることは、任意の数値 nについて、⼗分な回数のテストを⾏って nがすべ
てのテストをパスすることが確認できれば、この素数判定が間違う確率を好き
なだけ⼩さくできるということです。
残念ながら、この主張はあまり正しくありません。フェルマーテストを騙

してしまう数というものは存在します。素数でない数値 nで、a < nであるす
modulo m を求めるためには x modulo mと y modulo mを別々に求め、これらをかけ
算し、その結果の法mに関する剰余を求めればいいという事実に基づいています。例え
ば、eが偶数であれば be/2 modulo mを求め、それを⼆乗し、法mに関する剰余を取り
ます。このテクニックを使うと、計算する際に mよりずっと⼤きな数値を扱わないで済
ませられるので便利です。(練習問題 1.25と⽐較すること)

べての整数 a について、an が nを法として aに合同となってしまうものです。
そのような数は⾮常にまれなので、フェルマーテストは実⽤的にはかなり信頼
できます。47

騙されないフェルマーテストの変種も複数あります。それらのテストでは、
フェルマーテストと同様に、整数 nの素数判定を⾏うのに a < nである整数 a

を選び、nと aに依存するある条件についてチェックします (練習問題 1.28で、
そのようなテストの例が出てきます)。⼀⽅、フェルマーテストと違って、任意
の nに対し、nが素数でない限り a < nであるほとんどの整数 aについてその
条件が成⽴しないことが証明できます。そのため、もし nがランダムに選んだ
数 aについてテストを通過するなら、nが素数である確率は半々より⾼くなり
ます。もし nがランダムに選んだ aの値⼆つに対してそのテストを通過するな
ら、nが素数である確率は 4分の 3より⾼くなります。このテストをランダム
に選んだ数 aについて繰り返していくと、間違いの確率を好きなだけ低くでき
ます。
間違いの確率が好きなだけ⼩さくなることが証明できるテストの存在は、

この種類のアルゴリズムへの関⼼を引き起こしました。今では、そのようなア
ルゴリズムは 確率的アルゴリズム (probabilistic algorithm)と呼ばれるように
なっています。この領域では多くの研究が⾏われていて、確率的アルゴズムは
多くの領域に適⽤され、成果をもたらしてきました。48

練習問題 1.21:smallest-divisor⼿続きを使って次の数値の最⼩
の約数を求めよ。199, 1999, 19999.

47 フェルマーテストを騙してしまう数は カーマイケル数 (Carmichael number)と呼
ばれ、⾮常にまれであるということ以外はあまりよくわかっていません。100,000,000よ
り⼩さい数には 255のカーマイケル数が存在します。最⼩のものからいくつか挙げると、
561, 1105, 1729, 2465, 2821, 6601となっています。ランダムに選ばれた巨⼤数の素数性
をテストする際にたまたまフェルマーテストを騙す数値にぶつかってしまう確率は、“正
確な”アルゴリズムを実⾏する際に宇宙放射線がコンピュータのエラーを引き起こす確
率よりも低いのです。前者から考えるとアルゴリズムは不適切であると⾔える⼀⽅、後
者から考えると不適切ではないと⾔えます。これは数学と⼯学の違いを⽰しています。

48確率的素数判定の最も著しい応⽤は暗号分野です。現在、適当な 200桁の数値を因数
分解するというのは計算量的にできそうにありませんが、そのような数値の素数性チェ
ックはフェルマーテストによって数秒で⾏えます。この事実が、Rivest et al. (1977)に
よって提案された “解読不能な符号”を構築する技術の基礎となっています。その結果と
して⽣まれたRSAアルゴリズム (RSA algorithm)は、電⼦通信のセキュリティを強化
する技術として広く使われるようになっています。⼀時は、素数の研究というものは “純
粋”数学トピックの典型例で、研究のための研究のように思われていましたが、RSAア
ルゴリズムとその関連研究の進展によって、暗号学、電⼦送⾦、情報検索において重要
な実⽤的応⽤があることがわかりました。

練習問題 1.22: ほとんどの Lisp実装は runtimeという基本⼿続き
を持っていて、システムが動いた時間を整数 (例えば、マイクロ秒
で計測したもの)として返す。次の timed-prime-test ⼿続きは、
整数 nを引数として呼ばれると、nを表⽰し、nが素数であるかチ
ェックする。nが素数であれば、⼿続きは 3つのアスタリスクと、
テスト実⾏にかかった時間を表⽰する。
(define (timed-prime-test n)

(newline)
(display n)
(start-prime-test n (runtime)))

(define (start-prime-test n start-time)
(if (prime? n)

(report-prime (- (runtime) start-time))))
(define (report-prime elapsed-time)

(display " *** ")
(display elapsed-time))

この⼿続きを使って、指定した範囲の連続した奇数について素数
判定を⾏う⼿続き search-for-primesを書け。その⼿続きを使っ
て、1000, 10,000, 100,000 より⼤きな素数をそれぞれ 3つ⾒つけ
よ。判定アルゴリズムはΘ(

√
n)の増加オーダーを持っているので、

10,000あたりの判定には 1000あたりの √10倍程度の時間がかか
るはずである。あなたの計測データはこれを裏付けているだろう
か。100,000や 1,000,000のデータは、Θ(

√
n)という予想はどれだ

け当たっているだろうか。あなたの結果は、演算に必要なステッ
プ数に⽐例して実⾏時間が増えるという概念に⽭盾していないだ
ろうか。
練習問題 1.23: この節の最初で⽰した smallest-divisor⼿続き
は、必要のないテストを数多く⾏っている。数値が 2で割り切れ
るかチェックした後に、それより⼤きな偶数で割り切れるかをチ
ェックすることに意味はない。このことは、test-divisorに使わ
れる値は 2, 3, 4, 5, 6, . . . ではなく、2, 3, 5, 7, 9, . . .であるべきだ
ということを⽰している。この変更を実装するために、⼊⼒が 2
であれば 3を返し、そうでなければ⼊⼒に 2を⾜したものを返す
⼿続き nextを定義せよ。smallest-divisor⼿続きを、(+ test-
divisor 1) の代わりに (next test-divisor)を使うように修正

せよ。この変更したバージョンの smallest-divisor⼿続きを組
み込んだ timed-prime-testを使い、練習問題 1.22で⾒つけた 12
個の素数についてテストを実⾏せよ。この変更はテストのステッ
プ数を半分にするため、2倍の速度で実⾏できると予想できる。こ
の予想は確認できただろうか。そうでなければ、⼆つのアルゴリ
ズムの観測された速度の⽐率はどうなっていて、それが 2でない
という事実をどのように説明できるだろうか。
練習問題 1.24:練習問題 1.22の timed-prime-test⼿続きを fast-
prime?(フェルマー法)を使うように修正し、練習問題 1.22で⾒つ
けた 12 個の素数をそれぞれテストせよ。フェルマーテストは
Θ(logn)で増加するが、1,000,000に近い素数をテストするのにか
かる時間は 1000に近い素数をテストするのに必要な時間と⽐べて
どの程度になると予想できるか。実際とのずれがあれば、それを
説明できるだろうか。
練習問題 1.25: Alyssa P. Hackerは、expmodを書くのにだいぶ余
計な⼿間がかかったと愚痴をこぼした。どうせ指数の計算のやり
⽅はわかっているんだから、単純にこんなふうに書いたらよかっ
たんじゃないの、と。
(define (expmod base exp m)

(remainder (fast-expt base exp) m))

彼⼥は正しいだろうか。⾼速素数判定にこの⼿続きを使っても同
じようにうまく⾏くだろうか。
練習問題 1.26: Louis Reasonerは 練習問題 1.24を解くのにだいぶ
苦労している。彼の fast-prime? テストは、彼の prime?テストよ
りだいぶ遅いようだ。Louisは友達の Eva Lu Atorに⾒てもらうこ
とにした。⼆⼈で Louisのコードをよく⾒てみると、expmod⼿続
きで square⼿続きを呼ぶのでなく、明⽰的にかけ算を使うように
書き換えていることがわかった。
(define (expmod base exp m)

(cond ((= exp 0) 1)
((even? exp)
(remainder (* (expmod base (/ exp 2) m)

(expmod base (/ exp 2) m))

m))
(else
(remainder (* base

(expmod base (- exp 1) m))
m))))

Louisは、「何も違わないように⾒えるけど」と⾔っている。Eva
は、「いや、違うって。⼿続きの書き⽅のせいで、Θ(logn)のプロ
セスが Θ(n)のプロセスになっちゃったんだよ」と⾔う。
説明せよ。
練習問題 1.27: Footnote 1.47で挙げたカーマイケル数が実際にフ
ェルマーテストを騙すということを実験せよ。それはつまり、整
数 nを取り、すべての a < nについて an が法 nに関して aと合
同になるかテストする⼿続きを書き、与えられたカーマイケル数
に対してその⼿続きを試すということになる。
練習問題 1.28: 騙すことができないフェルマーテストの変種の
ひとつにミラー-ラビンテスト (Miller-Rabin test) (Miller 1976;
Rabin 1980)がある。これは、次のようなフェルマーの⼩定理の別
の形から始める。nが素数で、かつ aが n以下の任意の正の整数
であれば、aの (n − 1)乗は法 nに関して 1と合同であるという
ものだ。Miller-Rabinテストで数値 nの素数性をテストするには、
a < nである適当な aを選び、expmod⼿続きを使って aの (n− 1)

乗の nを法とする剰余を求める。しかし、expmod⼿続きの中で⼆
乗を⾏うステップで、毎回 “法 nに関する⾃明でない 1の平⽅根”
を⾒つけたかチェックする。これは、1とも n− 1とも等しくない
数値で、その⼆乗が法 nに関して 1に等しい数値である。そのよ
うな⾃明でない 1の平⽅根が存在すれば nは素数ではないという
ことは証明できる。また、もし nが素数でない奇数であれば、少な
くとも a < nである aのうち半分は、このように an−1を演算する
と、法 nに関する⾃明でない 1の平⽅根が現れるということも証
明できる (このため、Miller-Rabinテストは騙すことができない)。
expmod⼿続きを変更し、⾃明でない 1の平⽅根を⾒つけるとシグ
ナルを送るようにして、それを使って fermat-testと同じような
⼿続きとしてMiller-Rabinテストを実装せよ。既知の素数、⾮素
数を使ってその⼿続きをチェックせよ。ヒント：expmodにシグナ

ルを送らせる簡単な⽅法のひとつとして、0を返すようにするとい
うものがある。

1.3 ⾼階⼿続きによる抽象の定式化
ここまでで、⼿続きは事実上、数値に対する複合演算を特定の数値から独⽴し
た形で記述する抽象化であることを⾒てきました。例えば、次の⼿続きを⾒て
みましょう。
(define (cube x) (* x x x))

ここで扱っているのは、特定の値の三乗ではなく、任意の数値の三乗を得るや
り⽅です。もちろん、この⼿続きを定義しないでも、毎回次のような式を書く
こともできます。
(* 3 3 3)
(* x x x)
(* y y y)

こうすると、明⽰的に cubeについて書く必要はありません。しかし、このや
り⽅ではとても不利なことになります。たまたま⾔語が基本演算として持って
いた特定の演算 (この場合、かけ算)のレベルでずっと作業しなければならず、
⾼レベルの演算が使えないからです。プログラムは三乗を計算することはでき
ても、三乗という概念を表現する能⼒を⾔語内で持てないことになります。強
⼒なプログラミング⾔語であれば、よくあるパターンに名前をつけて抽象化し、
その抽象化したものを直接使って作業ができる能⼒が求められるところです。
⼿続きは、この能⼒を提供します。このため、極度に原始的なもの以外のすべ
てのプログラミング⾔語は、⼿続きを定義する仕組みを持っています。
しかし、数値計算であっても、もし引数が数値であるような⼿続きしか作れ

ないとしたら、抽象化の能⼒は⾮常に限定されたものになるでしょう。同じプ
ログラミングパターンがいくつもの異なる⼿続きに利⽤されるということはよ
くあります。そのようなパターンを概念として表現するには、⼿続きを引数と
して受け取ったり、⼿続きを返り値として返したりできる⼿続きを作れる必要
があります。⼿続きを操作する⼿続きは、⾼階⼿続き (higher-order procedure)
と呼ばれます。この節では、⾼階⼿続きが⾔語の表現⼒を⼤幅に⾼める強⼒な
抽象化メカニズムであることを⽰します。

1.3.1 引数としての⼿続き
次の 3つの⼿続きについて考えてみましょう。⼀つ⽬は、aから bまで整数の
和を計算するものです。
(define (sum-integers a b)

(if (> a b)
0
(+ a (sum-integers (+ a 1) b))))

⼆つ⽬は、範囲内の整数について、三乗の和を計算します。
(define (sum-cubes a b)

(if (> a b)
0
(+ (cube a)

(sum-cubes (+ a 1) b))))

三つ⽬は、次のような級数の和を計算します。
1

1 · 3 +
1

5 · 7 +
1

9 · 11 + . . . ,

これは、π/8に (⾮常にゆっくり)収束します。49

(define (pi-sum a b)
(if (> a b)

0
(+ (/ 1.0 (* a (+ a 2)))

(pi-sum (+ a 4) b))))

これらの三つの⼿続きの背後には、明らかに共通のパターンがあります。それ
らはほとんどの部分が同じもので、違うのは⼿続き名や、⾜し合わせる各項を
計算するために使われる aの関数、aの次の値を求める関数ぐらいです。それ
ぞれの⼿続きは、同じひな形の⽳を埋めることで⽣成できそうです。
(define (⟨name⟩ a b)

(if (> a b)
0

49この級数は Leibnizによるもので、普通は π
4
= 1− 1

3
+ 1

5
− 1

7
+ . . . という等価な形

で書かれます。3.5.3節では、これを基礎とした⾯⽩い数学上のトリックを紹介します。

(+ (⟨term⟩ a)
(⟨name⟩ (⟨next⟩ a) b))))

こういった共通パターンがあるということは、便利な抽象化が潜んでいて、⾒
つけ出されるのを待っているということの強⼒な証拠です。実際に、⼤昔の数
学者たちは 級数の総和 (summation of a series)という抽象化を発⾒し、“シグ
マ記法”を発明しました。例えば、次のような書き⽅でこの概念を表現します。

b∑
n=a

f(n) = f(a) + · · ·+ f(b),

シグマ記法の⼒によって、数学者は特定の値の総和についてだけでなく、総和
という概念そのものを扱えるようになりました。例えば、特定の級数から独⽴
して、総和についての⼀般的な結果について定式化できるようになったのです。
同様に、プログラムの設計者である私たちとしては、私たちの使う⾔語が

特定の総和を計算する⼿続きだけでなく、総和という概念そのものを表現する
⼿続きを書けるぐらいには強⼒であってほしいと思うところです。私たちの⼿
続き⾔語を使うと、上に書いた共通のひな形を持ってきて、その “⽳”を仮引数
に変えるだけですぐにできます。
(define (sum term a next b)

(if (> a b)
0
(+ (term a)

(sum term (next a) next b))))

sumが、下限と上限である a, bのほかに、⼿続き term(項), nextを取ることに
注意してください。sumは、ほかの⼿続きとまったく同じように使えます。例
えば、それを使って (引数に 1を⾜す⼿続き incも加えて)sum-cubesを定義で
きます。
(define (inc n) (+ n 1))
(define (sum-cubes a b)

(sum cube a inc b))

これを使うと、1から 10までの整数の三乗の和を計算できます。
(sum-cubes 1 10)
3025

項を計算するのに identity(恒等)⼿続きを使うと、sumによって sum-integers
を定義できます。
(define (identity x) x)
(define (sum-integers a b)

(sum identity a inc b))

こうすると、1から 10までの整数の合計が求められます。
(sum-integers 1 10)
55

pi-sumも、同じようにして定義できます。50

(define (pi-sum a b)
(define (pi-term x)

(/ 1.0 (* x (+ x 2))))
(define (pi-next x)

(+ x 4))
(sum pi-term a pi-next b))

これらの⼿続きを使うと、πの近似値が計算できます。
(* 8 (pi-sum 1 1000))
3.139592655589783

sumを⼿に⼊れたことで、それを構築⽤ブロックとして使って、また別の概念
を定式化できます。例えば、関数 f の範囲 aから bの定積分は、⼩さな dxの
値を使い、次の式によって数値的に近似できます。∫ b

a

f =

[
f

(
a+

dx

2

)
+ f

(
a+ dx+

dx

2

)
+ f

(
a+ 2dx+

dx

2

)
+ . . .

]
dx

これは、⼿続きとしてそのまま表現できます。
(define (integral f a b dx)

(define (add-dx x)
(+ x dx))

50ブロック構造 (1.1.8節)を使って pi-sumの中に pi-nextと pi-termの定義を埋め
込んでいることに注意してください。これらの⼿続きは、ほかの⼿続きの役には⽴ちそう
にないからです。この⼆つを厄介払いするやり⽅については、1.3.2節で⾒ていきます。

(* (sum f (+ a (/ dx 2.0)) add-dx b)
dx))

(integral cube 0 1 0.01)
.24998750000000042

(integral cube 0 1 0.001)
.249999875000001

(cubeを 0から 1の範囲で定積分したとき、その正確な値は 1/4です。)

練習問題 1.29: シンプソンの公式は、上で⽰したものより正確な
数値積分の⽅法である。シンプソンの公式では、関数 f の範囲 a

から bの定積分は次のように近似される。
h

3
(y0 + 4y1 + 2y2 + 4y3 + 2y4 + · · ·+ 2yn−2 + 4yn−1 + yn),

ここで、nを適当な偶整数として h = (b− a)/nで、yk = f(a+ kh)

である。(nを⼤きくすると近似の精度が上がる)。f , a, b, nを引
数に取り、シンプソンの公式によって計算した定積分の値を返す
⼿続きを定義せよ。その⼿続きを使い、(n = 100 と n = 1000で)
cubeの範囲 0から 1の定積分を求め、上で⽰した integral⼿続
きと結果を⽐較せよ。
練習問題 1.30:上で挙げた sum⼿続きは線形再帰を⽣成する。こ
の⼿続きは、総和計算が反復的に⾏われるように書き直すことが
できる。次の定義の中で⽋けている式を埋め、そのやり⽅を⽰せ。
(define (sum term a next b)

(define (iter a result)
(if ⟨??⟩

⟨??⟩
(iter ⟨??⟩ ⟨??⟩)))

(iter ⟨??⟩ ⟨??⟩))

練習問題 1.31:

63



a. sum⼿続きは、⾼階⼿続きとして表せる数多くの似たような
抽象化の中で、最も単純なものにすぎない。51特定範囲の点
における関数の値の積を返す似たような仕組みの⼿続きを、
productという名前で書け。productを使って factorialを
定義するやり⽅を⽰せ。また、次の公式52を使って、product
によって πの近似値を計算せよ。

π

4
=

2 · 4 · 4 · 6 · 6 · 8 · · ·
3 · 3 · 5 · 5 · 7 · 7 · · · .

b. あなたが書いた product ⼿続きが再帰プロセスを⽣成する
のであれば、線形プロセスを⽣成するものを書け。線形プロ
セスを⽣成するのであれば、再帰プロセスを⽣成するものを
書け。

練習問題 1.32:

a. sumと product (練習問題 1.31)は、どちらも accumulate(集
積) というより⼀般的な概念の特殊なケースであることを⽰
せ。集積とは、次のような⼀般的な集積関数を使って複数の
項を組み合わせるものである。
(accumulate combiner null-value term a next b)

accumulate は、sum や product と同じく項と範囲指定の引
数を取り、それに加えて、それまでの項の集積と現在の項を
どうやって結合するかを指定する combiner⼿続き (2引数)
と、項がなくなったときにどのような基本値を使うかを指定
する null-valueを引数に取る。accumulateを書き、sumと
productの両⽅を accumulateへの単純な呼び出しとして定
義できることを⽰せ。

51練習問題 1.31から練習問題 1.33は、たくさんの⼀⾒ばらばらの演算をまとめるため
に適切な抽象化を使うことで、どれだけの表現⼒が得られるかということを⽰すための
ものです。しかしながら、集積やフィルタというのは洗練された考え⽅ではあるのです
が、この時点でそれらを使うのは、両⼿を縛られているようなものです。これらの抽象
化を組み合わせる適切な⼿段となるデータ構造をまだ⼿に⼊れていないからです。2.2.3
節で、フィルタと集積を組み合わせるインターフェイスとして 列 (sequence)を使うやり
⽅を学んだのち、この考え⽅に戻ってきます。そこでは、プログラムを設計する強⼒で
洗練されたアプローチとしてこれらの⼿法が本領を発揮することになります。

52この公式は、17世紀のイギリス⼈数学者 John Wallisによって発⾒されたものです。

b. あなたが書いた accumulateが再帰プロセスを⽣成するので
あれば、線形プロセスを⽣成するものを書け。線形プロセス
を⽣成するのであれば、再帰プロセスを⽣成するものを書け。

練習問題 1.33: 結合する項に対するフィルタ (filter)という概念を
導⼊することで、さらに⼀般的なバージョンの accumulate (練習問
題 1.32)を⼿に⼊れることができる。指定された条件を満たす範囲
内の値から導出される項だけを結合するというものだ。結果となる
filtered-accumulate 抽象は、accumulate と同じ引数に加えて、
フィルタを指定する 1引数の述語を取る。filtered-accumulate
を⼿続きとして書け。次のものを filtered-accumulateを使って
どのように表現するかを⽰せ。

a. aから bの区間の素数の⼆乗の和 (すでに prime?述語を書い
ているとする)

b. n と互いに素である n 未満のすべての正の整数 (つまり、
gcd(i, n) = 1となるすべての整数 i < n)の積

1.3.2 lambdaを使って⼿続きを構築する
sumを使⽤する際に、1.3.1節でやったように、⾼階⼿続きの引数として使うた
めだけに pi-termや pi-nextのような些細な⼿続きを定義しないといけない
というのは、とんでもなく不格好に思えます。pi-nextや pi-termを定義する
より、“⼊⼒に 4を⾜したものを返す⼿続き”や “⼊⼒と⼊⼒に 2を⾜したもの
の積の逆数を返す⼿続き”を直接規定する⽅法があったほうが便利ではないで
しょうか。これは、⼿続きを作る lambdaという特殊形式を導⼊することでで
きるようになります。lambdaを使うと、やりたいことはそれぞれ次のように記
述できます。
(lambda (x) (+ x 4))

(lambda (x) (/ 1.0 (* x (+ x 2))))

こうすると、pi-sum⼿続きは次のように補助⼿続きをまったく定義しないで
表現できます。
(define (pi-sum a b)

(sum (lambda (x) (/ 1.0 (* x (+ x 2))))

a
(lambda (x) (+ x 4))
b))

同様に、lambdaを使うことで、integral⼿続きも補助⼿続き add-dxを定義し
ないで書けます。
(define (integral f a b dx)

(* (sum f
(+ a (/ dx 2.0))
(lambda (x) (+ x dx))
b)

dx))

⼀般的に、lambdaは define同様、⼿続きを作るために使われますが、その⼿
続きに名前を指定しないという点が違います。
(lambda (⟨formal-parameters⟩) ⟨body⟩)

作られる⼿続きは、defineによって作られるものと同じく、普通の⼿続きです。
唯⼀の違いは、環境の中で名前と関連づけられていないということだけです。
(define (plus4 x) (+ x 4))

これは、以下と同値です。
(define plus4 (lambda (x) (+ x 4)))

lambda式は、次のように読むことができます。訳注 2

(lambda (x) (+ x 4))
| | | | |

the procedure of an argument x that adds x and 4

値として⼿続きを持つほかの式と同じように、lambda式は次のように複合式
の中で演算⼦として使うことができます。
((lambda (x y z) (+ x y (square z)))
1 2 3)

12

訳注 2⽇本語では語順が異なるため、前から読み下すことはできません。意味は、「引数 x
を取り、xと 4を⾜す⼿続き」です。

66



というよりも、さらに⼀般的に⾔うと、通常⼿続き名を使うようなどんな⽂脈
でも使えます。53

letを使って局所変数を作る
lambdaの別の使い⽅として、局所変数を作るというものがあります。⼿続きの
中で、仮引数として束縛されたもの以外の局所変数を使いたいということはよ
くあることです。例えば、次の関数を計算したいとします。

f(x, y) = x(1 + xy)2 + y(1− y) + (1 + xy)(1− y),

これは、次のように表現することもできます。
a = 1 + xy,

b = 1− y,
f(x, y) = xa2 + yb+ ab.

f を計算する⼿続きを書く際には、xと yだけでなく、aや bといった計算過程
の値の名前も局所変数として持っておきたいところです。これを実現するひと
つの⽅法は、局所変数を束縛するために補助⼿続きを使うというものです。
(define (f x y)

(define (f-helper a b)
(+ (* x (square a))

(* y b)
(* a b)))

(f-helper (+ 1 (* x y))
(- 1 y)))

もちろん、lambda式を使って、局所変数を束縛するための無名⼿続きを指定す
ることもできます。そうすると、fの本体はその⼿続きの呼び出しひとつだけ
になります。

53lambdaという名前ではなく、make-procedure(⼿続き作成)のような当たり前の名前
をつけたほうが、Lispを勉強する⼈にわかりやすく、怖そうな感じを与えないですんだ
かもしれません。しかし、この慣習は深く根を下ろしたものです。この記法は、数理論
理学者の Alonzo Church (1941)によって導⼊された λ計算という数学的形式主義から
取られています。Churchは、関数と関数適⽤の概念を研究する厳密な基礎として λ計算
を開発しました。λ計算は、プログラミング⾔語の意味論を数学的に研究するための基礎
的な道具となっています。

(define (f x y)
((lambda (a b)

(+ (* x (square a))
(* y b)
(* a b)))

(+ 1 (* x y))
(- 1 y)))

この概念は便利なので、letという特殊形式でさらに⼿軽に使えるようになっ
ています。letを使うと、⼿続き fは次のように書けます。
(define (f x y)

(let ((a (+ 1 (* x y)))
(b (- 1 y)))

(+ (* x (square a))
(* y b)
(* a b))))

let式の⼀般形式は次のようになります。
(let ((⟨var1⟩ ⟨exp1⟩)

(⟨var2⟩ ⟨exp2⟩)
. . .

(⟨varn⟩ ⟨expn⟩))
⟨body⟩)

これは、次のように考えることができます。訳注 3

let ⟨var1⟩ have the value ⟨exp1⟩ and
⟨var2⟩ have the value ⟨exp2⟩ and
. . .

⟨varn⟩ have the value ⟨expn⟩
in ⟨body⟩

let式の最初の部分は、名前・式というペアのリストです。letが評価されると
き、それぞれの名前は対応する式の値と関連づけられます。letの本体は、こ
れらの名前が局所変数として束縛された状態で評価されます。これは、let式
が以下のものの別の⽂法として評価されているということです。
訳注 3ここでも、⽇本語では語順が異なります。「⟨body⟩の中で、⟨var1⟩には ⟨exp1⟩、⟨var2⟩
には ⟨exp2⟩…という値を持たせる」という意味です。

((lambda (⟨var1⟩ . . . ⟨varn⟩)
⟨body⟩)

⟨exp1⟩
. . .

⟨expn⟩)

局所変数を提供するのに、インタプリタの新しい仕組みは必要ありません。let
式は、その裏にある lambda適⽤に対するシンタックスシュガーにすぎません。
この等価関係から、let式で指定された変数のスコープが letの本体であ

るということがわかります。ここから、次のようなことを導くことができます。
• letを使うと、変数を可能な限り局所的に使⽤箇所に束縛できます。例
えば、もし xの値が 5であれば、次の式の値は 38になります。
(+ (let ((x 3))

(+ x (* x 10)))
x)

ここで、letの本体の中では xは 3なので let式の値は 33になります。
それに対して、⼀番外側の +の第⼆引数である xは 5のままです。

• 変数の値は、letの外側で計算されます。局所変数の値を提供する式が
局所変数⾃⾝と同じ名前を持った変数に依存している場合には、このこ
とが関係してきます。例えば、xの値が 2であれば、次の式の値は 12に
なります。
(let ((x 3)

(y (+ x 2)))
(* x y))

これは、letの本体の中では xは 3になり、yは 4(つまり、外側の xに 2
を⾜したもの)になっているからです。

letと同じ結果を得るために、内部定義が使える場合もあります。例えば、上
の⼿続き fは次のように定義することもできるところです。
(define (f x y)

(define a (+ 1 (* x y)))
(define b (- 1 y))
(+ (* x (square a))

(* y b)
(* a b)))

しかし、私たちはこのような状況では letを使い、defineは内部⼿続きの定義
に限定して使うようにしています。54

練習問題 1.34: 次の⼿続きを定義するとする。
(define (f g) (g 2))

すると、次のような結果になる。
(f square)
4
(f (lambda (z) (* z (+ z 1))))
6

もしここで、インタプリタに (f f)という組み合わせを評価させ
るという (ひねくれた) ことをすると、どのような結果が起こるだ
ろうか。

1.3.3 汎⽤⼿法としての⼿続き
1.1.4節では、数値計算のパターンを抽象化するメカニズムとしての複合⼿続
きを導⼊し、パターンを特定の数値から独⽴したものにしました。1.3.1節の
integral⼿続きのような⾼階⼿続きでは、さらに強⼒な抽象化に⾜を踏み⼊
れ、特定の関数から独⽴した計算の汎⽤⼿法を表現するために使われる⼿続き
というものを⾒てきました。この節では、もう⼆つの精巧な例、関数の零点と
不動点を⾒つける汎⽤⼿法について検討し、これらが⼿続きとして直接表現で
きるということを⽰します。

区間⼆分法によって⽅程式の根を求める
区間⼆分法 (half-interval method) とは、f という連続関数について⽅程式
f(x) = 0 の根を求めるための単純ながら強⼒なテクニックです。考え⽅は、
f(a) < 0 < f(b)となる点 aと bがあるとき、f は aと bの間に少なくともひと
つの零点を持つというものです。零点を特定するためには、aと bの平均を x

54内部定義を⼗分に理解し、望む通りの意味をプログラムに確実に持たせられるよう
になるには、この章で紹介したものよりも精巧な評価プロセスのモデルが必要です。し
かし、⼿続きの内部定義ではこの微妙な問題は起こりません。4.1.6節で評価についてよ
り深く学んだのちに、この問題に戻ってきます。

として f(x)を計算します。もし f(x) > 0なら、f は aと xの間に零点を持つ
ことになります。もし f(x) < 0なら、f は xと bの間に零点を持ちます。こ
の⽅法を続けると、f が零点を持つ区間をどんどん狭めていくことができます。
区間が⼗分狭くなったら、処理は停⽌します。不確定区間は処理の各ステップ
ごとに半分に縮まっていくので、必要なステップ数の増加オーダーは、元の区
間の⻑さを L、許容誤差 (“⼗分に⼩さい”と⾒なす区間の⼤きさ)を T として、
Θ(log(L/T ))になります。この戦略を実装した⼿続きを次に⽰します。
(define (search f neg-point pos-point)

(let ((midpoint (average neg-point pos-point)))
(if (close-enough? neg-point pos-point)

midpoint
(let ((test-value (f midpoint)))

(cond ((positive? test-value)
(search f neg-point midpoint))

((negative? test-value)
(search f midpoint pos-point))

(else midpoint))))))

関数 f と、それの値が負と正になる⼆つの点は、最初に与えられるものとしま
す。まず、⼆つの与えられた点の中間点を求めます。次に、与えられた区間が
⼗分に⼩さいかをチェックし、⼩さければ中間点を答えとして返します。そう
でない場合、中間点での f の値を計算し、それをテスト値とします。テスト値
が正であれば、元々の負の点から中間点までを新しい区間として、処理を続け
ます。もしテスト値が負であれば、中間点から正の点を区間として続けます。
最後に、テスト値が 0である可能性がありますが、その場合は中間点そのもの
が探索対象の根ということになります。
両端が “⼗分に近い”かどうかをテストするには、1.1.7節で平⽅根を求め

るために使ったのと同じような⼿続きが使えます。55

(define (close-enough? x y) (< (abs (- x y)) 0.001))

searchを直接使うと、ちょっと⾯倒なことがあります。うっかり f の値が符号
の条件を満たさないような点を与えてしまうかもしれないということです。そ

55ここでは、0.001を代表的な “⼩さな”値とし、計算の許容誤差を⽰すものとして使
っています。実際の計算における適切な許容誤差は、解くべき問題や、コンピュータや
アルゴリズムの制約によって変わります。これは⾮常に繊細な考慮が必要な問題となる
ことが多く、数値解析の専⾨家その他の魔法使いの助けが必要になることもあります。

の場合、間違った答えになってしまいます。そうする代わりに、以下に⽰す⼿
続きを経由して searchを使うことにしましょう。これは、どちらの端点が負の
関数値を持ち、どちらが正の関数値を持つかを調べ、それに応じて search⼿
続きを呼ぶというものです。関数が⼆つの与えられた点で同じ符号を持つ場合
は、区間⼆分法は使えませんので、⼿続きはエラーのシグナルを送ります。56

(define (half-interval-method f a b)
(let ((a-value (f a))

(b-value (f b)))
(cond ((and (negative? a-value) (positive? b-value))

(search f a b))
((and (negative? b-value) (positive? a-value))
(search f b a))

(else
(error "Values are not of opposite sign" a b)))))

次の例では、πを近似するのに、区間⼆分法を使って sinx = 0の 2と 4の間に
ある根として求めています。
(half-interval-method sin 2.0 4.0)
3.14111328125

もうひとつ例を挙げます。区間⼆分法を使って⽅程式 x3 − 2x − 3 = 0の 1か
ら 2の間の根を探すというものです。
(half-interval-method (lambda (x) (- (* x x x) (* 2 x) 3))

1.0
2.0)

1.89306640625

関数の不動点を求める
数値 xが⽅程式 f(x) = xを満たすとき、xは関数 f の 不動点 (fixed point)と
呼ばれます。関数 f によっては、最初の推定値から始めて、値があまり変わら
なくなるまで f を繰り返し適⽤していくというやり⽅で不動点を求めることが
できます。

f(x), f(f(x)), f(f(f(x))), . . . ,

56これは、ひとつの引数を取り、それをエラーメッセージとして表⽰する errorを使
って達成できます。

この考え⽅を使って、関数と初期推定値を⼊⼒として、その関数の不動点の近
似値を⽣成する⼿続きを作ることができます。前もって決めておいた許容範囲
に収まるような⼆つの値が得られるまで、関数を繰り返し適⽤していきます。
(define tolerance 0.00001)
(define (fixed-point f first-guess)

(define (close-enough? v1 v2)
(< (abs (- v1 v2))

tolerance))
(define (try guess)

(let ((next (f guess)))
(if (close-enough? guess next)

next
(try next))))

(try first-guess))

例えば、初期近似値の 1から始めてコサイン関数の不動点を近似するのにこの
⽅法が使えます。57

(fixed-point cos 1.0)
.7390822985224023

同様に、次の⽅程式の解を⾒つけることもできます。y = sin y + cos y:

(fixed-point (lambda (y) (+ (sin y) (cos y)))
1.0)

1.2587315962971173

不動点処理は、1.1.7節で平⽅根を求めるのに使った処理と似たところがあり
ます。どちらも、解が何らかの基準を満たすようになるまで繰り返し改善して
いくという考えに基づいています。実際に、平⽅根の計算は不動点探索として
すぐに形式化できます。ある数値 xの平⽅根を求めるには、y2 = xとなるよ
うな yを探す必要があります。この⽅程式を等価な形である y = x/yに変える
と、58 求めるものが y 7→ x/yの不動点であることがわかります。そのため、平
⽅根は次のような計算で求められないか試してみることができます。

57退屈な授業の間に、計算機をラジアンモードに設定し、不動点が得られるまで cos
ボタンを連打するということをやってみてください。

58 7→ (“maps to”(写像する)と読む)は、lambdaの数学者式の書き⽅です。y 7→ x/yは
(lambda (y) (/ x y)) という意味で、つまり yでの値が x/yとなるような関数を表し
ます。

(define (sqrt x)
(fixed-point (lambda (y) (/ x y))

1.0))

残念ながら、この不動点探索は収束しません。最初の推測値を y1とします。次
の推測値は y2 = x/y1 となり、その次は y3 = x/y2 = x/(x/y1) = y1 です。結
果として、y1 と y2 という⼆つの推測値をずっと繰り返し、答えの両側で振動
を続ける無限ループになります。
このような振動をコントロールする⽅法のひとつとして、推測値の⼤きな

変化を防ぐということがあります。答えは常に推測値 yと x/yの間にあるはず
なので、yと x/yの平均を取ることで、新しい推測値を yからも x/yからもあ
まり遠くないものにできます。つまり、y の次の推測値は 1

2
(y + x/y)となり

ます。
(define (sqrt x)

(fixed-point (lambda (y) (average y (/ x y)))
1.0))

(y = 1
2
(y + x/y)が⽅程式 y = x/y を単純に変形したものであることに注意し

てください。⽅程式の両辺に yを⾜し、2で割ることで得られます。)
この修正を加えると、平⽅根の⼿続きがうまくいくようになります。実は、

定義を外して中を⾒ていくと、ここで⽣成されている平⽅根の近似列が、1.1.7
節 の元の平⽅根⼿続きによって⽣成されるものと完全に同じであることがわ
かります。解に対する連続した近似値の平均を取るというこのアプローチは、
平均緩和法 (average damping)と呼ばれるテクニックで、不動点探索の収束に
役に⽴つことがよくあります。

練習問題 1.35: ⻩⾦⽐ φ(1.2.2節)が x 7→ 1 + 1/xという変形の不
動点であることを⽰し、そのことを利⽤して φを fixed-point ⼿
続きによって求めよ。
練習問題 1.36: fixed-point を修正し、練習問題 1.22で⽰した
newlineと displayという基本⼿続きを使って、⽣成する近似値
列を表⽰するようにせよ。次に、x 7→ log(1000)/ log(x)の不動点
を求めることによって xx = 1000の解を求めよ (⾃然対数を計算
する Scheme の基本⼿続き log を利⽤せよ)。平均緩和法を使う
場合と使わない場合のステップ数を⽐較せよ。(fixed-pointを開
始するのに、1という推定値が使えないことに注意。1を使うと、
log(1) = 0による割り算が発⽣してしまう。)

練習問題 1.37:

a. 無限連分数 (continued fraction)とは、以下のような形の式で
ある。

f =
N1

D1 +
N2

D2 +
N3

D3 + . . .

.

ひとつの例として、Ni とDi がすべて 1の無限連分数は 1/φ

になるということが⽰せる。ここで、φは (1.2.2節で説明し
た) ⻩⾦⽐である。無限連分数の近似値を求める⽅法のひと
つは、与えられた項数で展開を打ち切るというものである。
このように展開を打ち切ったもの— いわゆるk 項有限連分
数 (k-term finite continued fraction)—は、以下のような形に
なる。

N1

D1 +
N2

. . . +
Nk

Dk

.

nと dを 1引数 (項の添字 i)の⼿続きとし、それぞれ連分数
の項 Ni と Di を返すとする。(cont-frac n d k)を評価す
ると k項有限連分数の値を計算するような⼿続き cont-frac
を定義せよ。⼀連の kの値に対し、以下によって 1/φの近似
を求め、その⼿続きをチェックせよ。
(cont-frac (lambda (i) 1.0)

(lambda (i) 1.0)
k)

⼩数点以下 4桁の精度の近似を得るには、kをどれだけ⼤き
くしなければならないだろうか。

b. あなたが書いた cont-frac が再帰プロセスを⽣成するので
あれば、線形プロセスを⽣成するものを書け。線形プロセス
を⽣成するのであれば、再帰プロセスを⽣成するものを書け。

練習問題 1.38: スイスの数学者レオンハルト・オイラーは、1737
年に De Fractionibus Continuisという学術報告を発表した。それ

には、e− 2(eは⾃然対数の底)の連分数展開が含まれていた。この
分数では、Ni はすべて 1で、Di は 1, 2, 1, 1, 4, 1, 1, 6, 1, 1,. . .と
いう数列である。オイラーの連分数展開をもとに、練習問題 1.37の
cont-frac⼿続きを使って eを近似するプログラムを書け。
練習問題 1.39: 正接 (タンジェント) 関数の連分数表現は、1770年
にドイツの数学者ヨハン・ハインリッヒ・ランベルトによって発
表された。

tanx =
x

1−
x2

3−
x2

5− . . .

,

ここで、xはラジアンで表したものである。ランベルトの式をもと
に、正接関数の近似値を求める⼿続き (tan-cf x k)を定義せよ。
kは、練習問題 1.37同様、計算する項の数を指定する。

1.3.4 返り値としての⼿続き
ここまでの例では、⼿続きを引数として渡す能⼒が私たちのプログラミング⾔
語の表現⼒を⼤幅に拡張するということを⾒てきました。この表現⼒は、返り
値⾃⾝が⼿続きであるような⼿続きを作成することによって、さらに強⼒にで
きます。
この考えを理解するために、1.3.3節の最後で説明した不動点の例をもう⼀

度⾒てみましょう。この例では、√xは y 7→ x/y関数の不動点であるという観
察結果からスタートして、不動点探索という形で平⽅根⼿続きの新しいバージ
ョンを定式化しました。次に、平均緩和法を使って近似値が収束するようにし
ました。平均緩和法は、それ⾃体便利な汎⽤テクニックです。平均緩和法では、
関数 f が与えられたとき、それに対応する別の関数を考えます。その関数は、
xにおける値が xと f(x)の平均となるようなものです。
平均緩和法の考え⽅は、次の⼿続きによって表現できます。

(define (average-damp f)
(lambda (x) (average x (f x))))

average-dampは、引数として⼿続き fを取り、(lambdaで⽣成された) ⼿続き
を返り値とする⼿続きです。返り値となる⼿続きは、数値 xに適⽤すると xと
(f x)の平均を返すというものです。例えば、average-dampを square⼿続き

に適⽤すると、xでの値が xと x2の平均であるような⼿続きを返します。この
返された⼿続きに 10を適⽤すると、10と 100の平均、つまり 55を返します。
59

((average-damp square) 10)
55

average-dampを使うと、square-root⼿続きは次のように定式化し直すこと
ができます。
(define (sqrt x)

(fixed-point (average-damp (lambda (y) (/ x y)))
1.0))

この定式化は、この⼿法の三つの考え⽅を明確に⽰しています。不動点探索、
平均緩和法、そして関数 y 7→ x/yです。この平⽅根の⼿法の定式化を1.1.7節の
元のバージョンと⽐べると学ぶところがあるでしょう。これらの⼿続きは同じ
プロセスを表現しているということに気をつけて、これらの抽象化を使って表
現することでどれだけ考え⽅が明確になっているかを⾒てください。⼀般的に、
プロセスを⼿続きとして定式化するやり⽅はいくつもあります。経験豊富なプ
ログラマであれば、⼿続きによる定式化の⽅法から、特に明快で、プロセスの
要素が独⽴した実態としてほかのアプリケーションで再利⽤可能であるような
ものを選ぶことができます。再利⽤の単純な例として、xの三乗根が y 7→ x/y2

の不動点であり、そのため私たちの平⽅根の⼿続きはすぐに⽴⽅根を求めるも
のに拡張可能だということがあります。60

(define (cube-root x)
(fixed-point (average-damp (lambda (y) (/ x (square y))))

1.0))

ニュートン法
1.1.7節で初めて平⽅根の⼿続きを導⼊したとき、それが ニュートン法 (New-
ton’s method)の特別な場合だということに触れました。x 7→ g(x)が微分可能

59これは、演算⼦⾃⾝が複合式であるような複合式です。そのような複合式が作れると
いうことは、練習問題 1.4ですでに⾒ていますが、そこでの例はおもちゃのようなもので
した。ここからは、そういった複合式が実際に必要になるような場合—⾼階⼿続きの返
り値となる⼿続きを適⽤するとき—について⾒ていくことになります。

60さらなる⼀般化については、練習問題 1.45を参照してください。

な関数であるとき、⽅程式 g(x) = 0の解は、以下に⽰す関数 x 7→ f(x)の不動
点となります。

f(x) = x− g(x)

Dg(x)

ここで、Dg(x)は gの導関数を xで評価したものです。ニュートン法は、上で
⾒た不動点探索法を利⽤して、関数 f の不動点を探すことによって⽅程式の解
を近似するというものです。61

多くの関数 g と、⼗分によい初期推測値 x に対して、ニュートン法は
g(x) = 0の解に急速に収束します。62

ニュートン法を⼿続きとして実装するには、まず微分という考え⽅を表現
する必要があります。“微分”というものは、平均緩和法と同様に、ある関数
を別の関数に変形するものだということに注⽬してください。例えば、関数
x 7→ x3 を微分したものは x 7→ 3x2 です。
⼀般的に、gが関数で dxが⼩さな値であるとき、gを微分した Dgは、任

意の数値 xにおける値が (dxを極⼩にしたときに) 次のようになる関数です。

Dg(x) =
g(x+ dx)− g(x)

dx .

このため、微分という考え⽅は (例えば dxを 0.00001として)次のような⼿続
きとして表現できます。
(define (deriv g)

(lambda (x) (/ (- (g (+ x dx)) (g x)) dx)))

次の定義も加えます。
(define dx 0.00001)

average-damp同様、derivは⼿続きを引数に取り⼿続きを返り値とする⼿続
きです。例えば、x 7→ x3 を微分したものの x = 5での値 (正確な値は 75です)
の近似値を求めるには、次の式を評価します。

61微積分の⼊⾨書では、ニュートン法は普通、xn+1 = xn − g(xn)/Dg(xn)という近似
列として記述されています。私たちはプロセスに関して語る⾔語を持っているため、不
動点の考えを使って⼿法の記述を簡単にできます。

62ニュートン法は必ず解に収束するわけではありませんが、うまくいく場合は繰り返
しごとに解の近似値の精度の桁数が倍になるということが⽰せます。そのような場合は、
ニュートン法は区間⼆分法よりずっと速く収束します。

(define (cube x) (* x x x))
((deriv cube) 5)
75.00014999664018

derivの助けを借りると、ニュートン法は不動点探索プロセスとして表現
できます。
(define (newton-transform g)

(lambda (x) (- x (/ (g x) ((deriv g) x)))))
(define (newtons-method g guess)

(fixed-point (newton-transform g) guess))

newton-transform ⼿続きはこの節の最初の式を表現しており、newtons-
method はそれを使ってすぐに定義できます。引数としては、零点を⾒つけ
たい関数を計算する⼿続きと初期推測値を取ります。例えば、xの平⽅根は、初
期推測値 1から始めて、y 7→ y2 − x関数の零点をニュートン法によって求める
ことによって求められます。63

このことから、平⽅根⼿続きをまた別の形で書くことができます。
(define (sqrt x)

(newtons-method
(lambda (y) (- (square y) x)) 1.0))

抽象化とファーストクラス⼿続き
ここまで、平⽅根計算をより⼀般的な⼿法の具体例として表現する⼆つの⽅法
を⾒てきました。ひとつは不動点探索で、もうひとつはニュートン法です。ニ
ュートン法はそれ⾃⾝不動点プロセスとして表現されているので、実際には平
⽅根を不動点として計算する⼆つの⽅法を⾒てきたことになります。どちらも
ある関数から始め、その関数を変形したものの不動点を求めます。この⼀般的
な考え⽅⾃体、⼿続きとして表現できます。
(define (fixed-point-of-transform g transform guess)

(fixed-point (transform g) guess))

63平⽅根を求める場合、どんな初期値から初めても、ニュートン法は急速に正しい解に
収束します。

このとても⼀般的な⼿続きは、引数として⼿続き g(何らかの関数を計算す
る)、gを変形する⼿続き、初期推測値を取ります。返り値は変形された関数の
不動点です。
この抽象化によって、この節の最初の (y 7→ x/yの平均緩和バージョンに対

して不動点を求める)平⽅根計算を、この⼀般的な⽅法の⼀例として書き直す
ことができます。
(define (sqrt x)

(fixed-point-of-transform
(lambda (y) (/ x y)) average-damp 1.0))

同じようにして、この節の⼆つ⽬の (y 7→ y2 − xのニュートン変形の不動点を
求める)平⽅根計算は、次のように表現できます。
(define (sqrt x)

(fixed-point-of-transform
(lambda (y) (- (square y) x)) newton-transform 1.0))

1.3節は、複合⼿続きが⾮常に重要な抽象化メカニズムであるという観察結果
から始めました。複合⼿続きは、私たちのプログラミング⾔語の中で、⼀般的
な計算⼿法を明⽰的な要素として表現することを可能にするものだからです。
ここでは、⾼階⼿続きを使うことによって、これらの⼀般的な⼿法を組み合わ
せてさらなる抽象化を⽣み出していけるということを⾒てきました。
プログラマである私たちは、プログラムに潜んでいる抽象化を⾒つけ、そ

れらの抽象化を使ってプログラムを構築し、さらに強⼒な抽象化を⽣み出すた
めにそれらを⼀般化する機会を⾒逃さないようにしなければいけません。これ
は、いつでも可能な限り抽象的にプログラムを書かなければいけないというこ
とではありません。熟練プログラマであれば、タスクに対して適切な抽象レベ
ルを選べるものです。しかし、新しい状況でも適⽤できるように、このような
抽象化によって考えられるようになっておくことは重要です。⾼階⼿続きが重
要なのは、これらの抽象化を明⽰的に私たちのプログラミング⾔語で表現する
ことを可能にし、ほかの計算の要素と同じように扱えるようにしてくれるから
です。
⼀般的に、プログラミング⾔語は計算要素の操作⽅法に制約を課すもので

す。制約が最も少ない要素は、ファーストクラス (first-class)の地位を持つと⾔
います。ファーストクラス要素の “権利と特権”のいくつかを次に⽰します。64

64プログラミング⾔語の要素についてのファーストクラスの地位という概念は、イギ
リスの計算機科学者 Christopher Strachey (1916-1975)によるものです。

• 変数によって名前をつけることができる。
• ⼿続きに引数として渡すことができる。
• ⼿続きの返り値になることができる。
• データ構造に組み込むことができる。65

Lispは、ほかの⼀般的なプログラミング⾔語と違い、⼿続きに完全なファース
トクラスの地位を与えています。このことは効率的な実装をするうえでの課題
となりますが、そのおかげで⾮常に強い表現⼒を⼿にすることが可能です。66

練習問題 1.40: 次のような形の式で newtons-method⼿続きと組
み合わせて使うことによって、三次⽅程式 x3 + ax2 + bx + c = 0

の零点の近似値を求めることができるような⼿続き cubicを定義
せよ。
(newtons-method (cubic a b c) 1)

練習問題 1.41: 引数がひとつの⼿続きを引数として取り、その⼿
続きを⼆回適⽤する⼿続きを返す⼿続き doubleを定義せよ。例え
ば、incが引数に 1を⾜す⼿続きであれば、(double inc)は 2を
⾜す⼿続きになる。次の式はどんな値を返すだろうか。
(((double (double double)) inc) 5)

練習問題 1.42: f と g を⼆つの 1引数関数とする。g に f を合成
(composition)するということを、関数 x 7→ f(g(x))として定義す
る。合成を実装する⼿続き composeを定義せよ。例えば incが引
数に 1を⾜す⼿続きであれば、以下のようになる。
((compose square inc) 6)
49

65この実例は、第 2章でデータ構造を紹介した後に⾒ていきます。
66ファーストクラス⼿続きの実装コストのうち主要なものとして、⼿続きを返り値に
できるようにするには⼿続きの⾃由変数の領域を⼿続きが実⾏中でないときにも予約し
ておかなければいけないということがあります。4.1節で学ぶ Schemeの実装では、これ
らの変数は⼿続きの環境に保存されます。

練習問題 1.43: f が数値関数で nが正の整数であれば、f の n回適⽤
というものを作ることができ、それは xでの値が f(f(. . . (f(x)) . . . ))

である関数として定義できる。例えば、f が関数 x 7→ x+1であれ
ば、f の n回適⽤は関数 x 7→ x+ nとなる。もし f が数値を⼆乗
する演算であれば、f の n回適⽤は、引数を 2n乗する関数になる。
⼊⼒として f を計算する⼿続きと正の整数 nを取り、f の n回適
⽤を計算する⼿続きを返す⼿続きを書け。その⼿続きは、以下の
ように使えるはずである。
((repeated square 2) 5)
625

ヒント：練習問題 1.42の composeを使うと便利かもしれない。
練習問題 1.44: 関数の平滑化 (smoothing)は、信号処理において重
要な概念である。f が関数で dxがある⼩さな値であるとき、f を
平滑化したものは xでの値が f(x− dx), f(x), f(x+ dx)の平均と
なる関数である。⼊⼒として f を計算する⼿続きを取り、f を平
滑化したものを計算する⼿続きを返す⼿続き smoothを書け。時に
は、関数を繰り返し平滑化する (つまり、平滑化した関数をさら
に平滑化することを続ける)ことによって n 重平滑化関数 (n-fold
smoothed function)を得ることが役に⽴つこともある。smoothと練
習問題 1.43の repeated を使って任意の関数の n 重平滑化関数を
⽣成するやり⽅を⽰せ。
練習問題 1.45: 1.3.3節では、単純に y 7→ x/y の不動点を探すこ
とによって平⽅根を計算しようとすると収束しないということと、
それを平均緩和法によって解決できるということを⾒た。三乗根
の計算にも同じ⽅法を使うことができ、y 7→ x/y2 を平均緩和した
ものの不動点として求めることができる。残念ながら、このプロセ
スは四乗根ではうまくいかない。y 7→ x/y3 を収束させるのに、平
均緩和を⼀回するだけでは⼗分ではない。⼀⽅、平均緩和を⼆回⾏
う (つまり、y 7→ x/y3の平均緩和の平均緩和を使う)と、不動点探
索は収束するようになる。y 7→ x/yn−1を繰り返し平均緩和したも
のに対する不動点探索によって nth を計算する場合、何回の平均
緩和が必要となるか実験せよ。その結果を利⽤し、fixed-point,
average-damp および練習問題 1.43の repeated⼿続きを使って n

乗根を計算する単独の⼿続きを実装せよ。必要な数値演算は基本
⼿続きとして利⽤できるとする。

練習問題 1.46: この章で記述した数値計算法のいくつかは、反復
改良法 (iterative improvement)と呼ばれる⾮常に汎⽤的な計算戦
略の実例である。反復改良法というのは、何かを計算するにあた
って、答えに対する初期推測値から始めて、その推測値が⼗分によ
いかテストし、よくなければ推測値を改良し、改良した推測値を新
しい推測値としてそのプロセスを続けるというものである。⼆つ
の⼿続きを引数として取る⼿続き iterative-improveを書け。引
数はそれぞれ、推測値が⼗分によいか判断する⼿続きと、推測値
を改良する⼿続きとする。iterative-improveの返り値は、引数
として初期推測値を取り、それが⼗分によくなるまで改良を続け
る⼿続きとする。1.1.7節の sqrt⼿続きと1.3.3節の fixed-point
⼿続きを iterative-improveによって書き直せ。

2
データを⽤いた抽象化の構築

私たちはここで、数学的抽象化の決定的なステップに到達しまし
た。記号の意味について忘れるのです。. . .[数学者]は休む必要は
ありません。これらの記号を使ってできる演算はいくらでもあり
ます。記号の意味するものを⾒る必要さえありません。
—Hermann Weyl, The Mathematical Way of Thinking

第 1章では、計算プロセスと、プログラム設計における⼿続きの役割に焦点を
合わせました。基本データ (数値)と基本演算 (数値演算)の使い⽅、組み合わ
せや条件式や引数を使うことによって⼿続きを組み合わせて複合⼿続きを作る
⽅法、defineを使って⼿続きを抽象化する⽅法について⾒てきました。また、
プロセスの局所展開のパターンとして⼿続きを捉えることができるということ
も学びました。さらに、⼿続きという形で具体化されたプロセスのいくつかの
共通パターンについて分類や論証を⾏い、また簡単なアルゴリズム的解析をし
ました。また、⾼階⼿続きによって計算上の⼀般的⼿法を扱うことができるよ
うになり、またそれらを使って考えることができるようになるということ、そ
のことが私たちの⾔語を強⼒なものにしてくれるということも⾒てきました。
これは、プログラミングの本質の⼤きな部分を占めています。

この章では、さらに複雑なデータについて⾒ていきます。第⼀章での⼿続
きはすべて単純な数値データを処理するものでしたが、私たちがコンピュータ
で解決したいと思っている問題の多くでは、単純なデータは⼗分ではありませ
ん。プログラムは普通、複雑な現象をモデル化するように設計されるものです。
いくつもの側⾯を持つ実世界の現象をモデル化するために、いくつもの部品を
持つコンピュータ上のオブジェクトを作る必要に迫られることはよくありま
す。そのため、第⼀章での焦点が⼿続きを組み合わせて複合⼿続きを作ること
によって抽象化を構築することであったのに対して、この章では、どんなプロ
グラミング⾔語についても⾔える、もうひとつの重要な側⾯に⽬を向けます。
データオブジェクトを組み合わせて 複合データ (compound data)を作るため
にプログラミング⾔語が提供する⼿段です。
なぜプログラミング⾔語に複合データが必要なのでしょうか。その理由は、

複合⼿続きが必要である理由と同じです。プログラムを設計する概念レベルを
引き上げ、設計のモジュール性を⾼め、⾔語の表現⼒を強くしたいからです。
⼿続きを定義できることによって基本演算よりも⾼い概念レベルで⼿続きを扱
えるようになるように、複合データオブジェクトを構築できることによって、
⾔語の基本データオブジェクトよりも⾼い概念レベルでデータを扱えるように
なります。
有理数を使って計算を実⾏するシステムを設計するというタスクについて

考えてみましょう。例えば、⼆つの有理数を取って和を返す add-ratという演
算を考えることができます。単純なデータを使うと、有理数は分⺟と分⼦とい
う⼆つの整数として考えられます。そのため、有理数が⼆つの整数 (分⺟と分
⼦)として表され、add-ratが⼆つの⼿続き (ひとつは和の分⼦を返し、もうひ
とつは分⺟を返す) として実装されるようにプログラムを設計できます。しか
し、これは不格好です。こうすると、どの分⼦がどの分⺟に対応するのか明⽰的
に記録しておかないといけないからです。たくさんの有理数演算を実⾏するよ
うなシステムでは、そんな簿記みたいな細かい処理があるとプログラムがとて
も読みにくくなってしまいますし、⼈間の⼼に与える影響も⾔うまでもありま
せん。分⺟と分⼦を “くっつけて”、ペア—複合データオブジェクト (compound
data object)—にできれば、状況はずっとよくなります。そうすると、有理数を
ひとつの概念単位として捉えるという考え⽅に近いやり⽅でプログラムから扱
うことができます。
複合データを使うと、プログラムのモジュール性を上げることもできます。

有理数を⼀⼈前のオブジェクトとして直接操作できれば、プログラムの中で有
理数を有理数として扱う部分を、有理数を例えば整数のペアとしてどうやって
表すかといった細部から分離できます。プログラムの中のデータオブジェク

トをどうやって表すかを扱う部分と、データオブジェクトをどうやって使う
かを扱う部分とを分離するという汎⽤的なテクニックは、データ抽象化 (data
abstraction)と呼ばれる強⼒な設計⼿法です。ここでは、データ抽象化によっ
てプログラムの設計・保守・修正がどれだけ簡単になるかということを⾒てい
きます。
複合データを使うと、私たちのプログラミング⾔語の表現⼒を実際に向上

させることができます。“線形結合”ax+ byを作ることを考えてみましょう。書
き⽅はいろいろ考えられますが、例えば a, b, x, yを引数として取り、ax+ by

の値を返す⼿続きを書くとします。引数が数値であれば、何も難しいことはあ
りません。すぐに次のような⼿続きを定義できます。
(define (linear-combination a b x y)

(+ (* a x) (* b y)))

しかし、扱いたいものが数値だけでないとするとどうでしょう。加算と乗算さ
え定義されていれば、有理数でも複素数でも多項式でも何でも線形結合が作れ
るという考え⽅を、⼿続きという形で表現したいという場合について考えてみ
ます。これは、次のような形の⼿続きとして表現できます。
(define (linear-combination a b x y)

(add (mul a x) (mul b y)))

ここで addと mulは基本⼿続き +と *ではなく、a, b, x, yという引数とし
て渡すデータに対して、その種類に応じて適切な演算を⾏うという、より複雑
な何かです。ここでのポイントは、linear-combinationが a, b, x, yについて、
⼿続き addと mulがそれらに対して適切な演算を⾏ってくれるはずだというこ
とだけ知っていればいいというところです。⼿続き linear-combinationから
⾒ると、a, b, x, yが何であるかはどうでもいいことで、それらが基本データを
使ってどういう形で表現されることになっているかというのはもっとどうでも
いいことです。この例はまた、なぜ私たちのプログラミング⾔語が複合データ
を直接扱う能⼒を提供してくれないといけないのかを教えてくれます。それは、
その能⼒がないと、linear-combinationのような⼿続きが、内部構造を知ら
ないまま、引数を addと mulに渡すということを⾏う⽅法がないからです。1

1これは、⼿続きを直接操作する能⼒がプログラミング⾔語の表現⼒を向上させると
いうのと似たようなものです。例えば、1.3.1節では sumという⼿続きを導⼊しましたが、
これは⼿続き termを引数として取り、ある指定した区間での termの値の和を求めると
いうものでした。sumを定義するためには、termのような⼿続きについて、それがより
基本的な演算によってどのように表現されているかということと関係なく、独⽴した実

この章は、上で述べた有理数計算システムの実装から始めます。これによ
って、複合データやデータ抽象化について考察する基礎ができます。複合⼿続
きの場合と同じように、対象とする主な問題は、複雑性をうまく扱うテクニッ
クとしての抽象化です。ここでは、データ抽象化によって、プログラムの部品
間に適切な抽象化の壁 (abstraction barrier)を建てられるようになるというこ
とを⾒ていきます。
複合データを形成するうえで鍵となるのは、データオブジェクトを組み合

わせて複雑なデータオブジェクトを形成できるように、プログラミング⾔語が
ある種の “糊”を提供しなければいけないということです。糊として使うもの
には、いろいろな種類のものがありえます。その例として、特別な “データ”操
作をまったく使わないでも、⼿続きだけを使って複合データを作ることができ
るということを⾒ていきます。このことによって、第⼀章の末尾ですでに曖昧
になってきていた “⼿続き”と “データ”との境界線は、さらに曖昧なものにな
っていきます。また、列や⽊を表現するための確⽴されたテクニックについて
も⾒ていきます。複合データを扱ううえで鍵となる考え⽅のひとつは、クロー
ジャ (closure)という概念です。これは、私たちがデータオブジェクトを組み合
わせるのに使う糊は、基本データオブジェクトだけでなく、複合データオブジ
ェクトも組み合わせられるようになっていなければいけないということです。
もうひとつの鍵となる考え⽅は、部品を組み⽴てるようにプログラムのモジュ
ールを組み⽴てるうえで、複合データが標準インターフェイス (conventional
interface)として使えるということです。これらの考え⽅の⼀部については、ク
ロージャを使う簡単なグラフィック⾔語によって図解することにします。
また、記号式 (symbolic expression)—要素が数値だけでなく任意の記号か

らなるデータ—を導⼊することで、私たちの⾔語の表現⼒を強化します。さら
に、オブジェクトの集合を表現するうえでのいろいろなやり⽅について詳しく
⾒ていきます。ある数値関数を計算するのにいろいろな計算プロセスが考えら
れるのと同じように、あるデータ構造をより単純なオブジェクトによって表す
のにもいろいろなやり⽅があり、その表現をどう選ぶかということが、そのデ
ータを扱うプロセスが必要とする時間や空間に⼤きな影響を与えうるというこ
とがわかります。これらの考え⽅については、記号微分、集合の表現、情報の
エンコードといった⽂脈で検討していきます。
体として考えることができるということが決定的に重要です。もし “⼿続き”という概念
がなかったら、sumのような演算が定義できるという可能性を考えることさえできなか
ったかもしれません。それに、総和計算の実⾏ということに関する限り、termがより基
本的な演算によってどのように構築されているかという細かいところはどうでもいいこ
とです。

その次に、プログラムのいろいろな部分で異なる形で表現されているデー
タを扱うという問題を取り上げます。このためには、異なるデータ型を処理す
る ジェネリック演算 (generic operation)というものを実装しなければなりま
せん。ジェネリック演算というものがある状況でモジュール性を維持するには、
単純なデータ抽象化だけの場合に⽐べて、より強⼒な抽象化の壁が必要になり
ます。具体的には、データ主導プログラミング (data-directed programming)と
いうテクニックを導⼊します。これは、それぞれのデータ表現を独⽴して設計
し、それらを加法的 (additively)に (つまり、修正なしに)組み合わせられるよ
うにするものです。システム設計におけるこのアプローチの⼒を⽰すために、
そこまでで学んだことを多項式上での記号演算を実⾏するパッケージの実装に
適⽤することで、この章の締めくくりとします。ここでの多項式の係数は、整
数、有理数、複素数、さらにはほかの多項式となることもあります。

2.1 データ抽象化⼊⾨
1.1.8節では、ある⼿続きをより複雑な⼿続きの要素として使うとき、その⼿
続きは特定の演算の集合としてだけでなく、⼿続きの抽象化として⾒ることも
できるということに触れました。つまり、⼿続きがどうやって実装されている
かという細かいところは隠しておくことができ、その特定の⼿続き⾃⾝、同じ
ような全体的なふるまいを持つ別の任意の⼿続きによって置き換えることがで
きるということです。別の⾔い⽅をすると、⼿続きがどのように使われるかと
いうところを、その⼿続きがより基本的な⼿続きによってどのように実装され
ているかといった細かいところから分離するという抽象化ができるというこ
とになります。複合データについての同じような概念は、データ抽象化 (data
abstraction)と呼ばれます。データ抽象化は、複合データオブジェクトがどう
使われるかというところを、それがより基本的なデータオブジェクトによって
どのように構築されているかといった細かいところから分離することを可能に
する⽅法です。
データ抽象化の基本的な考え⽅は、複合データオブジェクトを使うような

プログラムを構築する際に、“抽象データ”を扱うようにするということです。
これは、プログラムでデータを扱うときに、⼿元のタスクを実⾏するのに絶
対に必要とは⾔えないような想定をしないようにするということです。また、
“具体的な”データ表現は、そのデータを使うプログラムとは独⽴に定義され
ます。私たちのシステムのこれら⼆つの部品をつなぐインターフェイスの役割
を担うのは⼀組の⼿続きで、それらは セレクタ (selector) とコンストラクタ

(constructor)と呼ばれます。これらが抽象的なデータを具体的な表現によって
実装することになります。このテクニックの説明のために、有理数を扱う⼿続
きの集合をどのように設計するかということを考えていきます。

2.1.1 例: 有理数の数値演算
有理数を使って数値演算を⾏いたいという場合について考えてみましょう。⾏
いたい演算は、⾜し算、引き算、かけ算、割り算や、⼆つの有理数が等しいか
どうかテストするといったものになるでしょう。
まずは、分⼦と分⺟から有理数を構築する⽅法はすでに持っていると仮定

しましょう。また、有理数が与えられたときに、その分⼦と分⺟を抽出 (セレ
クト)する⽅法もあるとします。さらに、コンストラクタとセレクタは⼿続き
として使うことができるとします。

• (make-rat ⟨n⟩ ⟨d⟩)は、分⼦が整数 ⟨n⟩で分⺟が整数 ⟨d⟩である有理数
を返す。

• (numer ⟨x⟩)は、有理数 ⟨x⟩の分⼦を返す。
• (denom ⟨x⟩)は、有理数 ⟨x⟩の分⺟を返す。

ここでは、プログラムを合成していくうえでの強⼒な戦略である 希望的思考
(wishful thinking)を使っています。有理数をどうやって表現するのか、numer,
denom, make-rat という⼿続きをどう実装するのか、まだ何も⾔っていません。
それでも、もしこれらの⼿続きを持っていたとすると、⾜し算、引き算、かけ
算、割り算、等価性のテストは、次のような関係を使って実⾏できます。

n1

d1
+
n2

d2
=
n1d2 + n2d1

d1d2
,

n1

d1
− n2

d2
=
n1d2 − n2d1

d1d2
,

n1

d1
· n2

d2
=
n1n2

d1d2
,

n1/d1
n2/d2

=
n1d2
d1n2

,

n1

d1
=
n2

d2
if and only if n1d2 = n2d1.

これらの規則は⼿続きとして表現できます。

(define (add-rat x y)
(make-rat (+ (* (numer x) (denom y))

(* (numer y) (denom x)))
(* (denom x) (denom y))))

(define (sub-rat x y)
(make-rat (- (* (numer x) (denom y))

(* (numer y) (denom x)))
(* (denom x) (denom y))))

(define (mul-rat x y)
(make-rat (* (numer x) (numer y))

(* (denom x) (denom y))))
(define (div-rat x y)

(make-rat (* (numer x) (denom y))
(* (denom x) (numer y))))

(define (equal-rat? x y)
(= (* (numer x) (denom y))

(* (numer y) (denom x))))

これで、有理数の演算が numer, denom, make-ratというセレクタ⼿続きと
コンストラクタ⼿続きによって定義できました。しかし、セレクタ⼿続きとコ
ンストラクタ⼿続きは定義していません。ここで、分⺟と分⼦を組み合わせて
有理数を作るために何らかの⽅法が必要になります。

ペア
テータ抽象化の具体的なレベルを実装することを可能にするために、私たちの
⾔語は ペア (pair)という複合構造を⽤意しています。これは、consという基
本⼿続きによって構築できます。この⼿続きは⼆つの引数を取り、それら⼆つ
の引数を部品として含む複合データオブジェクトを返します。ペアがあるとき、
それらの部品は carと cdrという基本⼿続きを使って取り出すことができま
す。2 そのため、cons, car, cdrは次のような使い⽅ができます。

2consという名前は “construct”を表しています。carと cdrという名前は、ibm 704
上での最初の Lisp実装に由来します。この機械はメモリ位置の “アドレス”と “デクリメ
ント”という部分が参照できるようなアドレス⽅式を持っていました。carは “Contents
of Address part of Register”(レジスタのアドレス部分の中⾝)を表し、cdr(“クダー”と
読みます)は “Contents of Decrement part of Register” (レジスタのデクリメント部分

(define x (cons 1 2))
(car x)
1
(cdr x)
2

ペアは、基本データオブジェクトと同じように、名前をつけて操作できるデー
タオブジェクトです。さらに、consは要素がペアであるペアや、またそれが要
素となるペアといったものを作ることもできます。
(define x (cons 1 2))
(define y (cons 3 4))
(define z (cons x y))
(car (car z))
1
(car (cdr z))
3

2.2節では、ペアを組み合わせるこの能⼒が、ペアというものをすべての複雑
なデータ構造を作るための汎⽤ブロックとして使えるものにしているというこ
とを学びます。唯⼀の基本複合データであるペア (pair)は、⼿続き cons, car,
cdrによって実装されていて、これが私たちが必要とするただひとつの糊です。
ペアによって構築されるデータオブジェクトは リスト構造 (list-structured)の
データと呼ばれます。

有理数を表現する
有理数システムを作るのに必要な残りの部分を補うために、ペアは⾃然なやり
⽅を提供してくれます。単純に、分⼦と分⺟という⼆つの整数のペアとして有
理数を表現すればいいのです。そうすると、make-rat, numer, denomはすぐに
次のように実装できます。

3

の中⾝)を表しています。
3セレクタとコンストラクタは次のように実装することもできます。

(define make-rat cons)
(define numer car)
(define denom cdr)

(define (make-rat n d) (cons n d))
(define (numer x) (car x))
(define (denom x) (cdr x))

また、計算結果を表⽰するために、有理数を分⼦、スラッシュ、分⺟として表
⽰することにします。4

(define (print-rat x)
(newline)
(display (numer x))
(display "/")
(display (denom x)))

これで、有理数⼿続きを試してみることができます。
(define one-half (make-rat 1 2))
(print-rat one-half)
1/2
(define one-third (make-rat 1 3))
(print-rat (add-rat one-half one-third))
5/6
(print-rat (mul-rat one-half one-third))
1/6
(print-rat (add-rat one-third one-third))
6/9

⼀つ⽬の定義は make-ratという名前を、式 consの値である、ペアを構築する基本⼿
続きに関連づけるというものです。こうすると、make-ratと consは同じ基本コンスト
ラクタに対する名前になります。
セレクタとコンストラクタをこのやり⽅で定義すると、効率はよくなります。make-

ratが consを呼び出すのではなく、make-ratは consであるため、make-ratが呼ばれ
るときに呼ばれる⼿続きは、⼆つではなくひとつになります。⼀⽅で、このやり⽅をし
てしまうと、⼿続き呼び出しを追跡したり⼿続き呼び出しにブレークポイントを置いた
りといったデバッグ⼿段が使い物にならなくなってしまいます。make-rat⼿続きの呼び
出しを監視したいという場合はあるかもしれませんが、consのすべての呼び出しを監視
するなんてことはしたくないはずです。
この本では、この定義スタイルは使わないことにしました。
4displayはデータを表⽰する Schemeの基本要素です。Schemeの基本要素 newline

は改⾏を⾏います。これらの⼿続きは、どちらも意味のある値は返しません。そのため、
以下での print-ratの使⽤例では、print-ratが表⽰するもののみを⽰し、インタプリ
タが print-ratの返り値として表⽰するものは⽰さないことにします。

最後の例からわかるように、この有理数計算の実装は有理数を既約のものに簡
約してくれません。これを直すには、make-ratを修正します。1.2.5節で扱っ
たような、⼆つの整数の最⼤公約数を返す gcd⼿続きがあれば、ペアを構築す
る前に gcdを使って分⼦と分⺟を既約にできます。
(define (make-rat n d)

(let ((g (gcd n d)))
(cons (/ n g) (/ d g))))

こうすることで、次のように望む結果が得られます。
(print-rat (add-rat one-third one-third))
2/3

この修正はコンストラクタ make-ratの変更だけで完了し、実際の演算を実装
する (add-ratや mul-ratといった)⼿続きはどれも変更する必要がありませ
んでした。

練習問題 2.1: 正と負の両⽅の引数を扱うことができる改良版
make-ratを定義せよ。make-ratは符号を正規化し、正の有理数で
あれば分⼦と分⺟の両⽅が正となり、負の有理数であれば分⼦の
みが負になるようにする。

2.1.2 抽象化の壁
複合データとデータ抽象化の例を続けて⾒ていく前に、有理数の例で出てきた
いくつかの問題について考えてみましょう。有理数演算を定義するのに、私た
ちはコンストラクタである make-ratとセレクタである numer、denomを使い
ました。⼀般的に、データ抽象化の底にある考え⽅は、それぞれのデータオブ
ジェクトの型に対して、それさえあればその型に対するどんな演算も⾏えるよ
うな基本演算セットを特定し、その後はデータを操作するのにそれらの演算し
か使わないようにするというものです。
有理数システムの構造は、図 2.1のようにイメージできます。⽔平線は、シ

ステムの異なる “レベル”を分離する抽象化の壁 (abstraction barrier)を表して
います。有理数を使うプログラムは、有理数パッケージによって “公共⽤”とし
て提供された⼿続きである add-rat, sub-rat, mul-rat, div-rat, equal-rat?
だけを使って有理数を操作します。これらの⼿続き⾃⾝はコンストラクタとセ
レクタである make-rat, numer, denomだけによって実装されていて、さらにコ
ンストラクタとセレクタ⾃⾝はペアによって実装されています。ペアは、cons,

有理数を使うプログラム

問題ドメインでの有理数

add-rat  sub-rat  ...

分母と分子としての有理数

make-rat  numer  denom

ペアとしての有理数

cons  car  cdr

ペアの実装（何でもよい）

図 2.1: 有理数パッケージ内の抽象化の壁

car, cdrによって操作することさえできれば、どのように実装されているかは
有理数パッケージの残りの部分とは無関係です。実質的に、それぞれのレベル
の⼿続きは、抽象化の壁を定義し、異なるレベルをつなぐインターフェイスと
なっています。
この単純な考え⽅には、いろいろないいところがあります。そのひとつは、

プログラムを保守したり修正したりするのが簡単になるということです。どん
な複雑な構造でも、プログラミング⾔語によって提供される基本データ構造を
使って、いろいろな⽅法で表すことができます。もちろん、どういう表し⽅を
選ぶかということは、それを扱うプログラムに影響を与えます。そのため、後
になってどこかの時点で表現⽅法を変えることになったら、それらのプログラ
ムはすべてそれに合わせて修正しないといけなくなるかもしれません。もし表
現⽅法への依存がごく少数のプログラムモジュールに制限されるように設計さ
れていなければ、⼤きなプログラムではこの作業は時間とコストがかかるもの
になるでしょう。
例えば、有理数を既約に簡約する問題の別の解き⽅として、簡約を実⾏す

るタイミングを、有理数の構築時でなく、有理数の部品へのアクセス時にする
というやり⽅があります。この場合、コンストラクタとセレクタは違ったもの
になります。

(define (make-rat n d) (cons n d))
(define (numer x)

(let ((g (gcd (car x) (cdr x))))
(/ (car x) g)))

(define (denom x)
(let ((g (gcd (car x) (cdr x))))

(/ (cdr x) g)))

この実装と前の実装との違いは、いつ gcdを計算するかというところです。プ
ログラムを普通に使う際に、同じ有理数の分⼦や分⺟に何回もアクセスするの
であれば、gcdは有理数の構築時に計算したほうがいいでしょう。そうでなけ
れば、gcdの計算はアクセス時まで遅らせたほうがいいかもしれません。どち
らを選ぶにしても、実装を切り替える場合に、add-rat, sub-ratなどはまった
く修正する必要がありません。
表現⽅法への依存を少しだけのインターフェイス⼿続きに限定することは、

プログラムの修正時だけでなく、プログラムの設計時にも役に⽴ちます。そう
しておくと、別の実装を考慮するための柔軟性を維持できるからです。上の単
純な例を引き続き使って、有理数パッケージを設計するにあたって、gcdを実
⾏するタイミングを構築時にするか選択時にするかを最初の段階で決められな
いとします。この場合、データ抽象化という⽅法を使うことによって、その決
定を後回しにしつつ、システムのほかの部分の開発を進めるということができ
るようになります。

練習問題 2.2:平⾯上の線分を表現するという問題について考える。
それぞれの線分は、始点と終点という点のペアとして表す。コンス
トラクタ make-segmentとセレクタ start-segment, end-segment
を定義せよ。さらに、点は x座標と y 座標という数値のペアとし
て表現できる。最後に、定義したセレクタとコンストラクタによ
って、線分を引数として取りその中点 (両端点の座標を平均した座
標を持つ点)を返す midpoint-segment ⼿続きを定義せよ。⼿続き
をチェックするためには、例えば次のようにして点を表⽰するこ
とが必要となるだろう。
(define (print-point p)

(newline)
(display "(")
(display (x-point p))
(display ",")

(display (y-point p))
(display ")"))

練習問題 2.3: 平⾯上の⻑⽅形の表現を実装せよ (ヒント：練習問
題 2.2を利⽤するといいかもしれない)。そのコンストラクタとセ
レクタを使って、ある⻑⽅形の外周の⻑さと⾯積を計算する⼿続
きを書け。次に、⻑⽅形の異なる表現を実装せよ。適切な抽象化の
壁を使ってシステムを設計し、外周と⾯積を求める同じ⼿続きが
どちらの表現によっても動くようにできるだろうか。

2.1.3 データとは何か
2.1.1節での有理数の実装では、有理数演算 add-rat, sub-rat 等を、make-rat,
numer, denomという仕様の決まっていない三つの⼿続きによって実装すること
から始めました。その時点では、それらの演算はデータオブジェクト—分⼦、
分⺟、有理数—によって定義され、データオブジェクトのふるまいは後者の三
つの⼿続きによって規定されると考えることができました。
しかし、データ (data)とは正確には何を意味しているのでしょうか。“ある

セレクタとコンストラクタによって実装される何か”と⾔うだけでは⼗分では
ありません。明らかに、任意の 3つの⼿続きのセットがすべて有理数の実装に
対する適切な基礎としての役割を果たせるわけではありません。有理数 xを n
と d という整数のペアによって構築する場合、xの numerと denomを取り出し
て割り算を⾏ったものが、nを dで割るのと同じ結果になるということを保証
する必要があります。別の⾔い⽅をすると、make-rat, numer, denomは、任意
の整数 nとゼロでない整数 dに対し、xが (make-rat n d)であるとき、次の
条件を満たす必要があります。

(numer x)
(denom x)

=
n
d
.

実際のところ、これは make-rat, numer, denomが有理数表現の基礎となる
ために満たさないといけないただひとつの条件です。⼀般的に、データという
ものは、何らかのセレクタとコンストラクタの集合に加え、それらが有効な表
現となるために満たさなければならないと規定された条件によって定義される
ものと考えることができます。5

5意外なことに、この考え⽅は厳密に定式化することが⾮常に難しいものです。そのよ
うな定式化には⼆つのアプローチがあります。ひとつは C. A. R. Hoare (1972) が開拓

この⾒⽅は、有理数のような “⾼レベル”のデータオブジェクトを定義す
る場合だけでなく、低レベルなオブジェクトに対しても使えます。有理数の定
義する際に使った、ペアという概念について考えてみましょう。ペアとは何か
ということについてはこれまで触れず、⾔語がペアを操作する⼿続きとして
cons, car, cdrを⽤意してくれているということだけを述べてきました。しか
し、これら三つの操作について知らなければいけないことは、⼆つのオブジェ
クトを consによってくっつけたら、それらのオブジェクトを carと cdrによ
って取得できるということだけです。つまり、それらの操作は、任意のオブジ
ェクト xと yに対して、もし zが (cons x y)であるなら (car z)は xであり、
(cdr z)は yであるという条件を満たすということです。これら三つの⼿続き
が私たちの⾔語に基本要素として含まれるということについてはすでに触れた
通りですが、上の条件を満たすものであれば、どんな三つの⼿続きであっても
ペアを実装する基礎として使うことができます。このことは、cons, car, cdr
を実装するときに、何のデータ構造も使わず、⼿続きだけによって⾏うことが
できるという事実が鮮やかに⽰してくれます。以下はその定義です。
(define (cons x y)

(define (dispatch m)
(cond ((= m 0) x)

((= m 1) y)
(else (error "Argument not 0 or 1: CONS" m))))

dispatch)
(define (car z) (z 0))
(define (cdr z) (z 1))

このような⼿続きの使い⽅は、データというものがどういうものかという私た
したもので、抽象モデル (abstract model)という⼿法として知られています。この⼿法
では、上の有理数の例で要点を述べた “⼿続きと条件”という仕様を定式化するものです。
有理数表現での条件は整数に関する事実 (等価性と割り算) によって記述されています。
⼀般的に、抽象モデルではそれまでに定義されたデータオブジェクトの型によって新し
い種類のデータオブジェクトを定義します。そのため、データオブジェクトに関する表
明は、それまでに定義済みのデータオブジェクトに関する表明に書き換えてチェックで
きます。もうひとつのアプローチは、mitの Zillesと ibmのGoguen, Thatcher, Wagner,
Wright (Thatcher et al. 1978参照)や、トロント⼤学のGuttag(Guttag 1977参照) によ
って導⼊されたもので、代数的仕様 (algebraic specification) と呼ばれます。代数的仕様
では、“⼿続き”を抽象代数システムの要素と⾒なし、そのふるまいは、ここでの “条件”
に相当する公理によって規定されるとし、データオブジェクトに関する表明には抽象代
数の技法を使います。どちらの技法も、Liskov and Zilles (1975)で調査されています。

ちの直感的な概念とまったく対応していません。それでも、これらの⼿続きが
上の条件を満たすということさえ確認できれば、これがペアを表現する有効な
やり⽅であるということが⽰せます。
ここで注意すべき微妙なところは、(cons x y)に返される値は⼿続き— 具

体的には、内部で定義された⼿続き dispatch—だということです。dispatch
は、ひとつの引数を取り、その引数が 0か 1かによって xか yかを返します。
それに対応して、(car z)は zに 0を適⽤するように定義されています。よっ
て、もし zが (cons x y)によって作られた⼿続きであれば、zを 0に適⽤する
と xを返します。こうして、(car (cons x y))が期待通り xを返すというこ
とが⽰せました。同様に、(cdr (cons x y))は (cons x y)の返り値となる⼿
続きを 1に適⽤し、yを返します。このため、この⼿続きによるペアの実装は
有効な実装であり、cons, car, cdrだけを使ってペアにアクセスするのであれ
ば、この実装を “本物の”データ構造を使う実装と区別することはできません。
⼿続きによるペアの実装の実例を⽰したことのポイントは、私たちの⾔語

がこういうふうに動いているというわけではなく (Schemeを含め、Lispシス
テムは⼀般的に、効率上の理由からペアを直接的に実装しています)、こういう
ふうに動くこともできるということです。⼿続きによる表現はわかりにくいで
すが、ペアが満たす必要のある条件を満たしているため、ペアを表現する完全
に適切な⽅法と⾔えます。ここでの例は、⼿続きをオブジェクトとして操作す
る能⼒は、⾃動的に複合データを表現する能⼒を提供するということも⽰して
います。これは、今の段階ではただの⾯⽩い考え⽅のように思えるかもしれま
せんが、⼿続きによるデータの表現というものは、私たちのプログラミングの
レパートリーの中で中⼼的な役割を果たしています。このプログラミングスタ
イルはよくメッセージパッシング (message passing)と呼ばれ、第 3章でモデ
ル化とシミュレーションの問題を扱う際には、これを基本的な道具として使う
ことになります。

練習問題 2.4: 次に⽰すものは、⼿続きによるペアの表現の別の
やり⽅である。この表現で、任意のオブジェクト xと yに対して、
(car (cons x y))が xを返すことを確認せよ。
(define (cons x y)

(lambda (m) (m x y)))
(define (car z)

(z (lambda (p q) p)))

対応する cdrの定義はどうなるだろうか。(ヒント：これが動作す
ることを確認するには1.1.5節の置換モデルを利⽤する)

98



練習問題 2.5: aと bのペアを積 2a3b の整数で表現することによっ
て、⾮負整数のペアを数値と数値演算だけを使って表現できると
いうことを⽰せ。
それに対応する cons, car, cdr⼿続きを定義せよ。
練習問題 2.6: ペアを⼿続きとして表現するという考え⽅で頭がご
ちゃごちゃになっていないとしたら、次のようなことを考えてみ
よう。⼿続きを扱うことができる⾔語では、(少なくとも、⾮負整
数に関する限りは)数値なしでもやっていける。そのためには、0
と、1を⾜すという演算を次のように実装する。
(define zero (lambda (f) (lambda (x) x)))
(define (add-1 n)

(lambda (f) (lambda (x) (f ((n f) x)))))

この表現は、発明者の Alonzo Church にちなんでチャーチ数
(Church numeral) として知られている。Alonzo Church は λ-演
算を発明した論理学者である。
oneと twoを直接 (zeroと add-1を使わずに)定義せよ (ヒント：
置換を使って (add-1 zero)を評価する)。加算⼿続きの直接的な
定義 +(add-1の繰り返し適⽤は⽤いない)を与えよ。

2.1.4 発展問題: 区間演算
Alyssa P. Hackerは⼯学上の問題の解決を助けるシステムを設計しています。
システムで提供したい機能のひとつとして、精度のわかっている不正確な量
(例えば物理装置の測定値)を扱う能⼒があります。そのような近似値を使って
計算を⾏った場合、結果が精度つきの数値となるようにしたいということです。
電気技師は Alyssaのシステムを使って電気に関するいろいろな値を計算し

ます。その際に、⼆つの抵抗 R1, R2の並列に等価な抵抗値 Rp を次の式によっ
て計算することが必要になるときがあります。

Rp =
1

1/R1 + 1/R2
.

抵抗値は普通、抵抗の製造者によって保証された何らかの許容誤差の範囲
内であることしかわかりません。例えば、“10% の許容誤差で 6.8Ω” という
ラベルのついた抵抗を買うと、その抵抗の抵抗値は 6.8 − 0.68 = 6.12 から

6.8 + 0.68 = 7.48Ωの間であるということしか確かには⾔えません。そのため、
6.8Ω 10%の抵抗と 4.7Ω 5%の抵抗を並列に接続すると、その組み合わせの抵
抗は約 2.58Ω(⼆つの抵抗が下限値である場合)から約 2.97Ω(⼆つの抵抗が上限
値である場合)までの範囲になります。

Alyssaのアイデアは、“区間”(不正確な量の取りうる値の範囲を表すオブジ
ェクト)を組み合わせる数値演算のセットとして “区間演算”を実装するという
ものでした。⼆つの区間を⾜し算、引き算、かけ算、割り算した結果は、それ
⾃⾝が区間であり、結果の範囲を表します。

Alyssaは、下限と上限という⼆つの端点を持つ “区間”という抽象オブジェ
クトがあると仮定しました。また、区間の両端点が与えられれば、データコン
ストラクタ make-intervalを使って区間を構築できるとしました。Alyssaは
まず、⼆つの区間を⾜す⼿続きを書きます。彼⼥は、和が取りうる最⼩値は⼆
つの下限の和であり、最⼤値は⼆つの上限の和だろうと推論します。
(define (add-interval x y)

(make-interval (+ (lower-bound x) (lower-bound y))
(+ (upper-bound x) (upper-bound y))))

また Alyssaは、両端点の積の最⼩値と最⼤値を求め、それらを答えとなる
区間の下限と上限とすることによって、⼆つの区間の積を計算しました (min
と maxは、任意の数の引数の最⼩値と最⼤値を⾒つける基本要素です)。
(define (mul-interval x y)

(let ((p1 (* (lower-bound x) (lower-bound y)))
(p2 (* (lower-bound x) (upper-bound y)))
(p3 (* (upper-bound x) (lower-bound y)))
(p4 (* (upper-bound x) (upper-bound y))))

(make-interval (min p1 p2 p3 p4)
(max p1 p2 p3 p4))))

⼆つの区間の割り算は、⼀つ⽬に⼆つ⽬の逆数をかけることにします。区間の
逆数の下限と上限は、上限の逆数と下限の逆数という順番になることに気をつ
けてください。
(define (div-interval x y)

(mul-interval
x
(make-interval (/ 1.0 (upper-bound y))

(/ 1.0 (lower-bound y)))))

練習問題 2.7: 区間の抽象化の実装を規定していないため、Alyssa
のプログラムは未完成である。区間のコンストラクタの定義は以
下のようになる。
(define (make-interval a b) (cons a b))

セレクタ upper-bound と lower-bound を定義し、実装を完成さ
せよ。
練習問題 2.8: Alyssaと同じような推論を使って、⼆つの区間の差
をどのように計算するかを記述せよ。それに対応する引き算の⼿
続き sub-intervalを定義せよ。
練習問題 2.9: 区間の幅 (width)は、上限と下限の差の半分である。
幅は、区間によって規定される数値の不確かさの程度である。い
くつかの数値演算では、⼆つの区間を組み合わせた結果の幅は、引
数の区間の幅のみによる関数となるが、それら以外の演算では、組
み合わせたものの幅は、引数の幅の関数とはならない。⼆つの区
間の和 (または差)の幅は、⾜される (または引かれる) 区間の幅の
関数となることを⽰せ。これが乗算と除算については正しくない
ということを例を挙げて⽰せ。
練習問題 2.10: エキスパートシステムプログラマの Ben Bitdiddle
は、Alyssaの肩越しにのぞいてきて、ゼロをまたぐ区間で割った
場合にどうなるかはっきりしていないとコメントした。Alyssaの
コードを修正し、この条件についてチェックして、もしこれが起
こればエラーのシグナルを送るようにせよ。
練習問題 2.11: ついでに Ben はまた謎めいたコメントを残した。
“区間の両端点の符号をテストすると mul-intervalは 9パターン
に場合分けできて、3回以上のかけ算が必要になるのはその中のひ
とつだけだよ。” Benの提案に従って、この⼿続きを書き直せ。

プログラムをデバッグしてから、Alyssaはユーザー候補の⼀⼈にそれを⾒せま
した。その⼈は、このプログラムは解いている問題が違う、ほしいのは中央値
と許容誤差で表される数値を扱うプログラムだと⾔います。例えば、3.5± 0.15

のような区間を使って作業をしたいのであって、[3.35, 3.65]ではないというこ
とです。Alyssaは⾃分のデスクに戻って、別のコンストラクタと別のセレクタ
を提供することによってこの問題を解決しました。

(define (make-center-width c w)
(make-interval (- c w) (+ c w)))

(define (center i)
(/ (+ (lower-bound i) (upper-bound i)) 2))

(define (width i)
(/ (- (upper-bound i) (lower-bound i)) 2))

運の悪いことに、Alyssaのユーザーのほとんどは技師です。実際の⼯学の現場
では、測定値の誤差は⼩さなもので、区間の中央値に対する幅の割合として表
されます。技師は普通、上で書いた抵抗のスペックのように、装置のパラメー
タに対するパーセント許容誤差を規定するものです。

練習問題 2.12:中央値とパーセント許容誤差を取り、求める範囲を
返すコンストラクタ make-center-percentを定義せよ。また、与え
られた区間に対してパーセント許容誤差を返すセレクタ percent
も定義する必要がある。centerセレクタは上に書いたものと同じ
である。
練習問題 2.13: パーセント許容誤差が⼩さいという前提のもとで
は、⼆つの区間の積のパーセント許容誤差を因数の許容誤差の積
によって近似する簡単な式が存在することを⽰せ。すべての数値
は正であると仮定して問題を単純化してもよい。

かなりの作業ののちに、Alyssa P. Hacker は完成したシステムを納品しまし
た。数年後、そのことをすっかり忘れていたころ、怒り狂ったユーザー Lem E.
Tweakitから興奮した⼝調の電話がかかってきました。どうも Lemは、並列抵
抗の式が代数的に等価な以下の⼆つの⽅法で書けることに気づいたようです。

R1R2

R1 +R2

1

1/R1 + 1/R2
.

彼は次の⼆つのプログラムを書きました。それぞれ、並列抵抗の式を異なった
やり⽅で計算します。
(define (par1 r1 r2)

(div-interval (mul-interval r1 r2)
(add-interval r1 r2)))

(define (par2 r1 r2)
(let ((one (make-interval 1 1)))

(div-interval
one (add-interval (div-interval one r1)

(div-interval one r2)))))

Lemは、Alyssaのプログラムがこれら⼆つの計算⽅法に対して異なる値を
返すと苦情を⾔っています。これは深刻な苦情です。

練習問題 2.14: Lemが正しいことを⽰せ。いろいろな数値演算の
式についてこのシステムの挙動を調査せよ。何らかの区間 Aと B

を作成し、それらを使って式 A/Aと A/B を計算せよ。中央値に
対して⼩さなパーセンテージの幅を持つ区間について調べると得
られるところが⼤きいだろう。中央値-パーセント形式 (練習問題
2.12参照)の計算結果を調べよ。
練習問題 2.15:別のユーザー Eva Lu Atorも、代数的に等価な異な
る式によって計算すると異なる区間になることに気がついた。彼
⼥は、不確定な数値を表す変数が繰り返し出てこないように書け
ば、より厳密な誤差限界を返すようにできると⾔う。つまり、並列
抵抗を計算するのに par2は par1よりも “よりよい” プログラム
だと⾔うのだ。彼⼥は正しいだろうか。また、それはなぜだろうか。
練習問題 2.16: ⼀般的に、なぜ代数的に等価な式が異なる答えに
なることがあるのかを説明せよ。この⽋点のない区間演算パッケ
ージを開発することはできるだろうか。それとも、このタスクは
不可能なのだろうか。(警告：この問題はとても難しい)

2.2 階層データと閉包性
ここまでで⾒てきたように、ペアは私たちが複合データオブジェクトを構築す
る基本的な “糊”となってくれます。図 2.2は、ペア—この場合は (cons 1 2)
で作られるペア—を図⽰する標準的なやり⽅を⽰しています。この表現は箱と
ポインタ記法 (box-and-pointer notation)と呼ばれるもので、それぞれのオブジ
ェクトは箱へのポインタ (pointer)として表されます。基本オブジェクトの箱
にはオブジェクトの表現が⼊っています。例えば、数値の箱には数字が⼊って

2

1

図 2.2: (cons 1 2)の箱とポインタ表現

1

4

2 3

3 4

1 2

(cons (cons 1 2)               (cons (cons 1
      (cons 3 4))                          (cons 2 3))
                                     4)

図 2.3: ペアを使って 1, 2, 3, 4を組み合わせる⼆つの⽅法

います。ペアの箱は実際には⼆つの箱で、左の箱にはペアの car(へのポイン
タ) が⼊り、右の箱には cdrが⼊ります。

consは、数値だけでなくペアも結合できるということについてはすでに⾒
ました (練習問題 2.2と練習問題 2.3を解く際には、この事実を使ったはずで
す)。結果として、ペアはどんな種類のデータ構造でも構築できる汎⽤ブロック
となってくれます。図 2.3はペアを使って数値 1, 2, 3, 4を組み合わせる⼆つの
⽅法を⽰しています。
ペアを要素とするペアを作る能⼒は、表現ツールとしてのリスト構造の持

つ重要性の本質となるものです。この能⼒のことを、cons の閉包性 (closure
property)と呼びます。⼀般的に、データオブジェクトを組み合わせる操作は、
その演算による組み合わせの結果⾃⾝が同じ演算によって組み合わせること
ができるのであれば、閉包性を満たします。6 閉包は、どんな組み合わせ⼿段

6ここでの “閉包”という⾔葉は抽象代数から来ています。抽象代数では、集合の要素

であっても、表現⼒の鍵となるものです。それによって、階層 (hierarchical)
構造—部品によって作られ、その部品もまた部品によって作られ . . .という構
造—を作ることが可能になるからです。
第 1章の初めから、⼿続きを扱う際には本質的に閉包を使ってきました。と

ても単純なプログラム以外は、組み合わせの要素はそれ⾃⾝組み合わせとなり
うるという事実に依存しているからです。この節では、複合データの閉包がも
たらす結果について扱います。また、ペアを使って列や⽊を表現するいくつか
の確⽴されたテクニックについて説明し、閉包を鮮やかに形にしてくれるグラ
フィック⾔語を紹介します。7

2.2.1 列の表現
ペアを使って構築できる便利な構造のひとつは、列 (sequence)— データオブジ
ェクトの順序つき集合—です。もちろん、ペアを使って列を表すやり⽅はいろ
いろあります。図 2.4には、特に素直な表現⽅法のひとつを⽰しています。この
やり⽅では、1, 2, 3, 4という列はペアのチェーンとして表されます。それぞれ
のペアの carはチェーン内の対応する項で、ペアの cdrはチェーン内の次のペ
アです。最後のペアの cdrは、ペアでない特別な値を指すことによって、列の
終端であることを伝えます。その値は、箱-ポインタ図では斜線として表され、
に演算を適⽤して得られるものがまたその集合の要素であるなら、それらの要素の集合
はその演算のもとで閉じていると⾔います。ところで、Lispコミュニティでは、(不幸な
ことに)“閉包”という⾔葉をまったく関係のない別の概念の記述にも使っています。その
場合の閉包とは、⾃由変数を持つ⼿続きを表現する実装上のテクニックを指しています。
この本では、そちらの意味での “閉包”は使いません。

7組み合わせ⼿段は閉包性を満たすべきだという概念は素直な考え⽅です。残念なこ
とに、⼀般的に使われる⾔語の多くでは、データ結合⼿段として提供されているものが
閉包性を満たさなかったり、閉包を使うのが⾯倒だったりします。Fortranや BASICで
は、普通は配列にまとめて⼊れることでデータを組み合わせます。しかし、要素がそれ
⾃⾝配列であるような配列は作れません。Pascalと Cでは、要素が構造体であるような
構造体を認めています。しかし、これをするためにはプログラマが明⽰的にポインタを
扱わなければならず、またプログラマは構造体の各フィールドが前もって決められた形
の要素しか持つことができないという制約に従わなければなりません。これらの⾔語は、
Lispがペアとして提供しているような、複合データを統⼀的な⽅法で簡単に操作できる
ビルトインの汎⽤の糊を持っていません。この制約が、Alan Perlisがこの本の前書きで
コメントしたことの背景にあります。“Pascalでは、宣⾔可能なデータ構造の多さが関数
の専⾨化を引き起こし、気軽な連携は禁じられ、ペナルティを与えられるものとなりま
す。ひとつのデータ構造を扱う 100の関数があるほうが、10のデータ構造を扱う 10の
関数があるのよりもよいのです。”

1 42 3

図 2.4: ペアの鎖として表現された列 1, 2, 3, 4

プログラムでは変数 nilという値として表されます。列全体は、⼊れ⼦の cons
演算によって構築されます。
(cons 1

(cons 2
(cons 3

(cons 4 nil))))

⼊れ⼦の consによって作られるこのようなペアの列はリスト (list)と呼ば
れます。Schemeは、リスト構築を便利にする listという基本⼿続きを提供し
ています。8 上の列は、(list 1 2 3 4)によって⽣成できます。⼀般的に、
(list ⟨a1⟩ ⟨a2⟩ . . . ⟨an⟩)

は次のものと等価です。
(cons ⟨a1⟩

(cons ⟨a2⟩
(cons . . .

(cons ⟨an⟩
nil). . .)))

Lispシステムは、慣例的にリストを括弧でくくられた要素の列として表⽰しま
す。そのため、図 2.4のデータオブジェクトは (1 2 3 4)と表⽰されます。
(define one-through-four (list 1 2 3 4))
one-through-four
(1 2 3 4)

8この本では、リスト (list)という⾔葉はリスト終端マーカーで終わるペアのチェーン
という意味で使います。それに対して、リスト構造 (list structure)という⽤語は、リス
トに限らず、ペアによって作られた任意のデータ構造を指します。

式 (list 1 2 3 4) と、その式を評価した結果として得られるリスト
(1 2 3 4)を混同しないように気をつけてください。式 (1 2 3 4) を評価
しようとすると、インタプリタは⼿続き 1を引数 2, 3, 4に適⽤しようとして、
エラーを起こします。

carはリストの最初の項を選択し、cdrは最初の項以外のすべてからなるサ
ブリストを選択するものであると考えることもできます。carと cdrを⼊れ⼦
で適⽤していくと、リストの⼆つ⽬、三つ⽬、それからそれに続く項を取り出
すことができます。9 コンストラクタ consは、元のリストと同じリストに最初
にひとつ項が加わったものを作ります。
(car one-through-four)
1
(cdr one-through-four)
(2 3 4)
(car (cdr one-through-four))
2
(cons 10 one-through-four)
(10 1 2 3 4)
(cons 5 one-through-four)
(5 1 2 3 4)

nilという値はペアのチェーンを終了させるために使うものですが、これを要
素のない列空リスト (empty list)と考えることもできます。nil という単語は、
“無”を意味するラテン語の単語 nihilを短くしたものです。10

9carと cdrの⼊れ⼦適⽤は書くのが⾯倒なので、Lispの諸⽅⾔はそれらの省略形を
提供しています。例えば、
(cadr ⟨arg⟩) = (car (cdr ⟨arg⟩))

このような⼿続きの名前はすべて cで始まり rで終わります。間の aは car演算を表
し、dは cdr演算を表し、名前の中での順番と同じ順番で適⽤されます。cadrのような
単純な組み合わせが発⾳可能になるように、carと cdrの名前を残しています。

10Lispの諸⽅⾔の標準化にあたって、どれだけのエネルギーが⽂字通り無に消えたか
というのは特筆すべきことです。nilは普通の名前であるべきか、nilの値は記号である
べきか、リストであるべきか、ペアであるべきか、といったものです。Schemeでは nil
は普通の名前で、この節では値がリスト終端マーカーであるような変数として使います
(trueが真の値を持つ普通の変数であるのと同じです)。Common Lispなどの Lisp⽅⾔
では、nilを特別な記号として扱います。この本の著者たちは⾔語標準化で多くの争いを
経験してきたので、この問題そのものを避けたいと思っています。2.3節でクォートを導
⼊したら、空リストを '()として⽰し、完全に変数 nilを使わないですむようにします。

リスト演算
ペアを使って要素の列をリストとして表現するやり⽅は、リストを “cdrダウ
ン”していくという確⽴されたプログラミングテクニックとセットになってい
ます。例えば、list-ref⼿続きは引数としてリストと数値 nを取り、リストの
n番⽬のものを返します。リストの要素の番号は 0から始めるのが慣例となっ
ています。list-refを計算する⽅法は以下の通りです。

• n = 0であれば、list-refはリストの carを返す。
• そうでなければ、list-refはリストの cdrの (n− 1)番⽬のものを返す。

(define (list-ref items n)
(if (= n 0)

(car items)
(list-ref (cdr items) (- n 1))))

(define squares (list 1 4 9 16 25))
(list-ref squares 3)
16

リスト全体を cdrダウンすることもよくあります。それをやりやすくするため
に Schemeは基本述語 null?を持っています。これは、引数が空リストかどう
かテストするものです。lengthというリストの要素数を返す⼿続きは、その典
型的な使⽤パターンを⽰しています。
(define (length items)

(if (null? items)
0
(+ 1 (length (cdr items)))))

(define odds (list 1 3 5 7))
(length odds)
4

⼿続き lengthは、単純な再帰計画を実装しています。簡約ステップは以下の
通りです。

• 任意のリストの length は、リストの cdr の length に 1 を⾜した値で
ある。

これは、次の基本ケースに着くまで連続的に適⽤されます。

• 空リストの lengthは 0である。
lengthは、反復スタイルで計算することもできます。
(define (length items)

(define (length-iter a count)
(if (null? a)

count
(length-iter (cdr a) (+ 1 count))))

(length-iter items 0))

もうひとつの確⽴されたプログラミングテクニックは、リストを cdrダウン
しながら答えのリストを “cons アップ” するというものです。これは、次の
append ⼿続きで使われます。append⼿続きは、⼆つのリストを引数として取
り、要素を連結して新しいリストを作ります。
(append squares odds)
(1 4 9 16 25 1 3 5 7)
(append odds squares)
(1 3 5 7 1 4 9 16 25)

appendも再帰計画を使って実装されています。リスト list1と list2を append
するには、次のようにします。

• もし list1が空リストであれば、結果は単に list2である。
• そうでなければ、list1の cdrと list2を appendし、その結果に list1
の carを consする。

(define (append list1 list2)
(if (null? list1)

list2
(cons (car list1) (append (cdr list1) list2))))

練習問題 2.17: 与えられた (空でない) リストの最後の要素のみを
持つリストを返す⼿続き last-pairを定義せよ。
(last-pair (list 23 72 149 34))
(34)

練習問題 2.18:リストを引数として取り、同じ要素を逆順に持つ
リストを返す⼿続き reverseを定義せよ。
(reverse (list 1 4 9 16 25))
(25 16 9 4 1)

練習問題 2.19: 1.2.2 節の両替パターン数計算プログラムについ
て考える。プログラムで使う通貨を変えられるようにして、例え
ばイギリスポンドの両替パターン数を計算できるようになると便
利だろう。現状のプログラムでは、通貨についての知識は⼀部は
first-denominationに、⼀部は count-change (アメリカには 5種
類のコインがあるということを知っている)に分散されている。両
替に使うコインのリストを与えられるようになればもっとよくな
るだろう。
⼿続き ccを書き直して、⼆つ⽬の引数として取るものを、どのコ
インを使うかを指定する整数ではなく、使うコインの値のリスト
となるようにしたい。それから、それぞれの通貨の種類を定義す
るリストを次のように持っておく。
(define us-coins (list 50 25 10 5 1))
(define uk-coins (list 100 50 20 10 5 2 1 0.5))

そして、ccは次のように呼び出すことにしよう。
(cc 100 us-coins)
292

このためには、プログラム ccに対して何らかの変更を加える必要
がある。プログラムの形は同じだが、⼆つ⽬の引数に対するアク
セスの仕⽅が次のように変わる。
(define (cc amount coin-values)

(cond ((= amount 0) 1)
((or (< amount 0) (no-more? coin-values)) 0)
(else
(+ (cc amount

(except-first-denomination
coin-values))

(cc (- amount

(first-denomination
coin-values))

coin-values)))))

リスト構造に対する基本演算によって、⼿続き first-denomination,
except-first-denomination, no-more?を定義せよ。リスト coin-
valuesの順序は、ccによって返される解答に影響を与えるだろう
か、それとも与えないだろうか。それはどうしてだろうか。
練習問題 2.20: ⼿続き +, *, listは任意の数の引数を取る。この
ような⼿続きを定義する⽅法のひとつとして、defineでドット末
尾記法 (dotted-tail notation)を使うというものがある。⼿続きの
定義の中で、仮引数リストの最後の仮引数名の前にドットのある
ものは、⼿続きが呼ばれるときに、前のほうの仮引数は (もしあれ
ば)通常通り前のほうの引数の値を持つことになるが、最後の仮引
数の値は残りの引数すべてのリスト (list)となる。例えば、次のよ
うな定義の場合、
(define (f x y . z) ⟨body⟩)

⼿続き fは、⼆つ以上の引数で呼び出すことができる。次のもの
を評価すると、
(f 1 2 3 4 5 6)

fの本体では、xは 1となり、yは 2で、zはリスト(3 4 5 6)とな
る。次の定義の場合、
(define (g . w) ⟨body⟩)

⼿続き gは 0個以上の引数で呼び出すことができる。次のものを
評価すると、
(g 1 2 3 4 5 6)

gの本体では、wはリスト (1 2 3 4 5 6)となる。11

11lambdaを使って fと gを定義する場合、次のように書けます。
(define f (lambda (x y . z) ⟨body⟩))
(define g (lambda w ⟨body⟩))

same-parityは、ひとつ以上の整数を引数として取り、最初の引数
と同じ偶奇性を持つ引数すべてのリストを返す⼿続きとする。例
えば、次のようになる。
(same-parity 1 2 3 4 5 6 7)
(1 3 5 7)
(same-parity 2 3 4 5 6 7)
(2 4 6)

上記の記法を使ってこの⼿続きを書け。

リストに対するマップ
⾮常に便利な演算として、リストのそれぞれの要素に何らかの変換を適⽤し、
結果のリストを返すというものがあります。例えば、次の⼿続きは与えられた
係数をリストのそれぞれの数値にかけます。
(define (scale-list items factor)

(if (null? items)
nil
(cons (* (car items) factor)

(scale-list (cdr items)
factor))))

(scale-list (list 1 2 3 4 5) 10)
(10 20 30 40 50)

1.3節でやったように、この⼀般的な考え⽅を抽象化し、⾼階⼿続きによって
表現される共通パターンとして捉えることができます。ここでの⾼階⼿続きは、
mapと呼ばれるものです。mapは、1引数の⼿続きとリストを引数として取り、
リストのそれぞれの要素に⼿続きを適⽤した結果のリストを返します。12

12 Schemeは、ここで記述したものよりも汎⽤的な map⼿続きを標準で提供していま
す。そのより汎⽤的な mapは、引数として n引数の⼿続きと n個のリストを取り、⼿続
きを、すべてのリストの⼀つ⽬の要素、すべてのリストの⼆つ⽬の要素 . . .に対して適⽤
していき、結果のリストを返します。例えば、次のようになります。
(map + (list 1 2 3) (list 40 50 60) (list 700 800 900))
(741 852 963)
(map (lambda (x y) (+ x (* 2 y)))

(list 1 2 3)

(define (map proc items)
(if (null? items)

nil
(cons (proc (car items))

(map proc (cdr items)))))
(map abs (list -10 2.5 -11.6 17))
(10 2.5 11.6 17)
(map (lambda (x) (* x x)) (list 1 2 3 4))
(1 4 9 16)

これで、mapを使って scale-listの新しい定義を書くことができます。
(define (scale-list items factor)

(map (lambda (x) (* x factor))
items))

mapは重要な概念ですが、それは mapが共通パターンを捉えているという理由
からだけではありません。mapはリストを扱ううえでのより⾼いレベルの抽象
化を達成するからです。元の scale-listの定義では、プログラムは再帰構造
になっていて、リストの要素ごとの処理に注意が向くようになっていました。
mapによって scale-listを定義すると、そのレベルの細かいことは隠されて、
係数倍というものが要素のリストを結果のリストに変換するものであるという
ところが強調されます。⼆つの定義の違いは、コンピュータが実⾏するプロセ
スの違いではなく (プロセスは同じものです)、私たちがプロセスをどう捉える
かという違いです。実際に mapは、リストを変形する⼿続きの実装を、リスト
の要素をどうやって取り出して結合していくかという細かいところから切り離
す抽象化の壁を作ることを助けてくれます。図 2.1で⽰した壁のように、この
抽象化を使うことで、列を列に変形するという概念的な演算の枠組みを保った
ままで、列の実装⽅法という低レベルな詳細を変更できる柔軟性が得られます。
2.2.3節では、列をプログラム組織化の枠組みとするというこの使い⽅につい
てさらに詳しく説明します。

練習問題 2.21: ⼿続き square-listは、引数として数値のリスト
を取り、それらの数値の⼆乗の列を持つリストを返す。
(list 4 5 6))

(9 12 15)

(square-list (list 1 2 3 4))
(1 4 9 16)

次に⼆つの異なる square-listの定義を⽰す。⽋けた部分を埋め、
⼆つとも完成させよ。
(define (square-list items)

(if (null? items)
nil
(cons ⟨??⟩ ⟨??⟩)))

(define (square-list items)
(map ⟨??⟩ ⟨??⟩))

練習問題 2.22: Louis Reasonerは練習問題 2.21の⼀つ⽬の square-
list⼿続きを書き直し、反復プロセスを展開するようにしようと
している。
(define (square-list items)

(define (iter things answer)
(if (null? things)

answer
(iter (cdr things)

(cons (square (car things))
answer))))

(iter items nil))

残念ながら、このように square-listを定義すると、答えとなる
リストは望むものの逆順になってしまう。なぜだろうか。
それから、Louisは consの引数を逆順にしてバグを直そうとした。
(define (square-list items)

(define (iter things answer)
(if (null? things)

answer
(iter (cdr things)

(cons answer
(square (car things))))))

(iter items nil))

これもうまくいかない。説明せよ。
練習問題 2.23: ⼿続き for-eachは mapに似ている。引数として⼿
続きと要素のリストを取る。しかし、結果のリストを⽣成するの
ではなく、for-eachは⼿続きをそれぞれの要素に左から右に順番
に⼿続きを適⽤していくだけである。⼿続きを要素に適⽤して返
される値はまったく使わない—for-eachは、表⽰のような、何か
のアクションを実⾏する⼿続きとともに使う。例えば次の例で、
(for-each (lambda (x)

(newline)
(display x))

(list 57 321 88))
57
321
88

for-eachへの呼び出し (上では⽰していないが)に返される値は
何でもよく、例えば真の値などでもよい。for-eachを実装せよ。

2.2.2 階層構造
リストによる列の表現は、要素がそれ⾃⾝列であるような列を表現するように
⾃然に⼀般化できます。例えば、次の式によって構築されるオブジェクト ((1
2) 3 4)は、
(cons (list 1 2) (list 3 4))

三つの項⽬を持つリストとして⾒ることができ、その⼀つ⽬の項⽬はそれ⾃⾝
が (1 2)というリストということになります。実際、このことはインタプリタ
によって表⽰される結果の形からも⾒て取ることができます。図 2.5は、この
構造のペアによる表現を⽰しています。
要素が列であるような列は、⽊ (tree)として考えることもできます。列の

要素は⽊の枝で、それ⾃⾝が列である要素は部分⽊となります。図 2.6に、図
2.5を⽊として⾒た構造を⽰します。
再帰は⽊構造を扱う⾃然なツールです。再帰を使うことで、⽊への演算を

その枝に対する演算に縮約し、それを今度は枝の枝に対する演算に縮約し . . .

と続けていくことで⽊の葉にたどり着く、というようにできることがよくあ

(1 2)
4

1 2

3

(3 4)

((1 2) 3 4)

図 2.5: (cons (list 1 2) (list 3 4)) によって作られ
る構造

((1 2) 3 4)

(1 2)

3 4

1 2

図 2.6: 図 2.5のリスト構造を⽊として⾒たもの

ります。ひとつの例として、2.2.1節の length⼿続きと、⽊の葉の総数を返す
count-leaves⼿続きを⽐べてみましょう。
(define x (cons (list 1 2) (list 3 4)))
(length x)
3
(count-leaves x)
4
(list x x)
(((1 2) 3 4) ((1 2) 3 4))
(length (list x x))
2

(count-leaves (list x x))
8

count-leavesを実装するために、lengthの計算で使った再帰計画を思い出し
てみましょう。

• リスト xの lengthは、xの cdrの lengthに 1を⾜した値である。
• 空リストの lengthは 0である。

count-leavesも似たようなものになります。空リストの場合の値は同じです。
• 空リストの count-leavesは 0である。

しかし、リストの carを取り外す縮約ステップでは、その car⾃⾝が⽊であっ
て、その葉の数も数えなければいけないかもしれないということを考える必要
があります。そのため、適切な縮約ステップは次のようになります。

• ⽊ xの count-leavesは、xの carの count-leavesと xの cdrの count-
leavesを⾜した値である。

最後に、carを取っていくと実際の葉に到達するので、基本ケースがもうひと
つ必要になります。

• 葉の count-leavesは 1である。
⽊に対する再帰⼿続きが書きやすくなるように、Schemeは pair?という基本
述語を提供しています。これは、引数がペアであるかどうかをテストするもの
です。以下に完成した⼿続きを書きます。13

(define (count-leaves x)
(cond ((null? x) 0)

((not (pair? x)) 1)
(else (+ (count-leaves (car x))

(count-leaves (cdr x))))))

練習問題 2.24: 式 (list 1 (list 2 (list 3 4)))を評価すると
する。インタプリタの表⽰する結果、それに対応する箱-点構造、⽊
としての解釈 (図 2.6で⽰したようなもの)を⽰せ。

13condの⼀番⽬と⼆番⽬の節は、この順番でなければいけません。空リストは null?
を満たし、またペアでもないからです。

練習問題 2.25: 以下のそれぞれのリストから 7を取り出す carと
cdrの組み合わせを書け。
(1 3 (5 7) 9)
((7))
(1 (2 (3 (4 (5 (6 7))))))

練習問題 2.26: ⼆つのリスト xと yを定義するとする。
(define x (list 1 2 3))
(define y (list 4 5 6))

次のそれぞれの式を評価すると、インタプリタはどのような結果
を表⽰するだろうか。
(append x y)
(cons x y)
(list x y)

練習問題 2.27: 練習問題 2.18の reverse⼿続きを修正し、deep-
reverseという⼿続きを書け。deep-reverseは、ひとつのリスト
を引数として取り、要素が逆順で、サブリストもすべて要素が逆
順になっているリストを返す⼿続きである。例えば、次のように
なる。
(define x (list (list 1 2) (list 3 4)))
x
((1 2) (3 4))
(reverse x)
((3 4) (1 2))
(deep-reverse x)
((4 3) (2 1))

練習問題 2.28: リストとして表現された⽊を引数として取り、そ
の⽊のすべての葉を左から右の順で要素として持つリストを返す
⼿続き fringeを書け。例えば、次のようになる。
(define x (list (list 1 2) (list 3 4)))
(fringe x)

(1 2 3 4)
(fringe (list x x))
(1 2 3 4 1 2 3 4)

練習問題 2.29: ⼆枝モビールは、左の枝と右の枝のという⼆つの
枝から構成されている。それぞれの枝はある⻑さを持つ棒で、おも
りか別の⼆枝モビールがぶら下がっている。⼆枝モビールは、⼆
つの枝によって構成することで、(例えば listを使って) 複合デー
タによって表現できる。
(define (make-mobile left right)

(list left right))

枝は、⻑さ length(必ず数値)と構造 structureからなり、structure
は数値 (単純なおもりを表す)か、または別のモビールである。
(define (make-branch length structure)

(list length structure))

a. これに対応する、モビールの枝を返すセレクタ left-branch
と right-branch、枝の構成要素を返す branch-length と
branch-structureを書け。

b. これらのセレクタを使って、モビールの総重量を返す⼿続き
total-weightを定義せよ。

c. モビールについて、バランスが取れている (balanced)という
のは、⼀番上の左枝にかかるトルク (回転⼒)が⼀番上の右の
枝にかかるトルクと等しく (つまり、左の棒の⻑さとかかる
重さの積が、右の棒についてのその積と等しい) 、かつ両枝
からぶら下がっている各部分モビールのバランスが取れてい
る状態について⾔う。ある⼆枝モビールがバランスが取れて
いるかどうかテストする述語を設計せよ。

d. モビールの表現を変更し、以下のようなコンストラクタに
する。
(define (make-mobile left right) (cons left right))
(define (make-branch length structure)

(cons length structure))

あなたのプログラムを新しい表現に移⾏するにはどの程度の
変更が必要だろうか。

⽊に対するマップ
mapは列を扱うための強⼒な抽象化ですが、それと同じように、mapに再帰を
組み合わせたものは⽊を扱うための強⼒な抽象化になります。例えば、scale-
treeという⼿続きは、2.2.1節の scale-listと似たようなものですが、引数
としては、数値の係数と、葉が数値である⽊を取ります。返り値は同じ形の⽊
で、それぞれの数値が係数倍されたものとなります。scale-treeの再帰計画
は、count-leavesのものに似ています。
(define (scale-tree tree factor)

(cond ((null? tree) nil)
((not (pair? tree)) (* tree factor))
(else (cons (scale-tree (car tree) factor)

(scale-tree (cdr tree) factor)))))
(scale-tree (list 1 (list 2 (list 3 4) 5) (list 6 7)) 10)
(10 (20 (30 40) 50) (60 70))

scale-treeの別の実装⽅法として、⽊を部分⽊の列と⾒なして mapを使う
というものがあります。列にマップを⾏い、それぞれの部分⽊を順番に係数倍
し、結果のリストを返すということになります。⽊が葉である基本ケースでは、
単純に係数倍します。
(define (scale-tree tree factor)

(map (lambda (sub-tree)
(if (pair? sub-tree)

(scale-tree sub-tree factor)
(* sub-tree factor)))

tree))

⽊の操作の多くは、列の操作と再帰を組み合わせて同じように実装できます。
練習問題 2.30: 練習問題 2.21の square-listと似たような⼿続き
square-treeを定義せよ。square-treeは以下のように動作する。
(square-tree
(list 1

(list 2 (list 3 4) 5)
(list 6 7)))

(1 (4 (9 16) 25) (36 49))

直接的な (つまり、⾼階⼿続きをまったく使わない)やり⽅と、map
と再帰を使うやり⽅の両⽅で、square-treeを定義せよ。
練習問題 2.31: 練習問題 2.30の解答を抽象化し、⼿続き tree-map
を作れ。square-treeが tree-mapによって以下のように定義でき
るようにせよ。
(define (square-tree tree) (tree-map square tree))

練習問題 2.32: 集合は、それぞれ異なる要素を持つリストとして
表現できる。また、集合のすべての部分集合の集合は、リストのリ
ストとして表現できる。例えば、集合が (1 2 3)の場合、すべて
の部分集合の集合は (() (3) (2) (2 3) (1) (1 3) (1 2) (1 2
3))となる。以下に、集合のすべての部分集合の集合を⽣成する⼿
続きを⽰す。これを完成させ、なぜそれが動作するのかを明確に
説明せよ。
(define (subsets s)

(if (null? s)
(list nil)
(let ((rest (subsets (cdr s))))

(append rest (map ⟨??⟩ rest)))))

2.2.3 標準インターフェイスとしての列
ここまで複合データについて⾒てきましたが、その中では、データ抽象化のお
かげでデータ表現の細かいところに煩わされることなくプログラムが設計で
きるということ、また抽象化によってほかの表現を実験する柔軟性を確保でき
るということを強調してきました。この節では、データ構造を扱ううえでの
もうひとつの強⼒な設計原則を導⼊します。それは、標準インターフェイス
(conventional interface)の使⽤です。

1.3節では、数値を扱うプログラムについて、プログラムの抽象化を⾼階⼿
続きとして実装することによって共通のパターンを捉えることができるという
ことを⾒てきました。複合データについて同じような演算を定式化する能⼒は、

私たちがデータ構造を操作するスタイルに決定的に依存します。例えば、次の
⼿続きについて考えてみます。これは2.2.2節の count-leavesと同じような仕
組みで、⽊を引数に取り、奇数の葉の⼆乗の合計を計算するというものです。
(define (sum-odd-squares tree)

(cond ((null? tree) 0)
((not (pair? tree))
(if (odd? tree) (square tree) 0))

(else (+ (sum-odd-squares (car tree))
(sum-odd-squares (cdr tree))))))

表⾯的には、この⼿続きは以下に⽰す⼿続きとまったく違うように⾒えます。
以下の⼿続きは、与えられた整数 n以下の kについて、フィボナッチ数 Fib(k)
のうち偶数のもののリストを構築するというものです。
(define (even-fibs n)

(define (next k)
(if (> k n)

nil
(let ((f (fib k)))

(if (even? f)
(cons f (next (+ k 1)))
(next (+ k 1))))))

(next 0))

これら⼆つの⼿続きは構造的には⼤きく違いますが、⼆つの計算をより抽象的
に記述すると、とても似通っていることが明らかになります。最初のプログラ
ムがしていることは、以下のように書くことができます。

• ⽊の葉を列挙し、
• フィルタによって奇数を選び、
• 選ばれた数の⼆乗を求め、
• +を使って、0から始めて結果を集積する。

⼆つ⽬のプログラムは、以下の通りです。
• 0から nまでの数値を列挙し、
• それぞれの整数に対するフィボナッチ数を求め、

• フィルタによって偶数を選び、
• consを使って、空リストから始めて結果を集積する。

信号処理のエンジニアであれば、図 2.7に⽰したように、直列につながれたい
くつかの段階を通る信号の流れとしてこれらのプロセスを概念化するのが⾃
然だと思うでしょう。sum-odd-squaresは 列挙 (enumerator)から始まります。
この列挙は、与えられた⽊が持つ葉からなる “信号”を⽣成します。この信号
は フィルタ (filter)に渡され、奇数要素以外がすべて取り除かれます。その結
果となる信号は、今度はマップ (map)に渡されます。このマップは、それぞれ
の要素に squareを適⽤する “変換器”です。マップの出⼒は、それから集積器
(accumulator)に与えられ、集積器は 0から始めて要素を +によって組み合わ
せます。even-fibsの図式も、同じような仕組みです。
残念ながら、上に⽰した⼆つの⼿続きの定義は、この信号の流れという構

造を⾒せることができていません。例えば、sum-odd-squaresを詳しく⾒てみ
ると、列挙の処理は、⼀部は null?と pair?というテストによって実装され、
⼀部は⽊の再帰という⼿続きの構造によって実装されています。同じように、
集積の処理の⼀部はテストのところにあり、⼀部は再帰の中で使われている⾜
し算のところにあります。全体的に、どちらの⼿続きにも、信号の流れという
描写の中の要素に対応するはっきりとした部分というものはありません。この
⼆つの⼿続きでは、計算過程を違う⽅法で分解しています。列挙処理をプログ
ラム全体に広げ、それをマップ、フィルタ、集積の処理と混ぜ合わせているの
です。信号の流れという構造が⼿続きの中で明確に⾒て取れるようにプログラ
ムを構成できれば、結果となるコードの概念的な明確さを⾼めることができる
でしょう。

列の演算
信号の流れという構造をより明確に反映するようにプログラムを構成するうえ
でのポイントは、処理の中で各段階の間を流れる “信号”に集中するというこ
とです。これらの信号をリストとして表現するなら、各段階の処理を実装する
のにリスト演算を使うことができます。例えば、信号の流れ図の中でのマップ
段階は、2.2.1節の map⼿続きを使って実装できます。
(map square (list 1 2 3 4 5))
(1 4 9 16 25)

列をフィルタリングして、与えられた述語を満⾜する要素だけを選択するのは、
次のようにすればできます。

enumerate:
tree leaves

filter:
odd?

map:
square

accumulate:
+, 0

enumerate:
integers

map:
fib

filter:
even?

accumulate:
cons, ()

図 2.7: ⼿続き sum-odd-squares(上)と even-fibs(下)を信
号の流れという図式によって表現すると、⼆つのプログラ
ムの共通性が明らかになる。

(define (filter predicate sequence)
(cond ((null? sequence) nil)

((predicate (car sequence))
(cons (car sequence)

(filter predicate (cdr sequence))))
(else (filter predicate (cdr sequence)))))

例えば、次のようになります。
(filter odd? (list 1 2 3 4 5))
(1 3 5)

集積は、次のように実装できます。
(define (accumulate op initial sequence)

(if (null? sequence)
initial
(op (car sequence)

(accumulate op initial (cdr sequence)))))
(accumulate + 0 (list 1 2 3 4 5))
15
(accumulate * 1 (list 1 2 3 4 5))
120
(accumulate cons nil (list 1 2 3 4 5))
(1 2 3 4 5)

後は、信号の流れ図を実装するのに⾜りないものは、処理する要素の列を列挙
することだけです。even-fibsの場合、⽣成するものは与えられた範囲の整数
列で、次のように実装できます。
(define (enumerate-interval low high)

(if (> low high)
nil
(cons low (enumerate-interval (+ low 1) high))))

(enumerate-interval 2 7)
(2 3 4 5 6 7)

⽊の葉の列挙は、次のようにします。14

(define (enumerate-tree tree)
(cond ((null? tree) nil)

((not (pair? tree)) (list tree))
(else (append (enumerate-tree (car tree))

(enumerate-tree (cdr tree))))))
(enumerate-tree (list 1 (list 2 (list 3 4)) 5))
(1 2 3 4 5)

これで、sum-odd-squaresと even-fibsを、信号の流れ図として⽰したような
形で定式化し直すことができます。sum-odd-squaresのほうは、⽊の葉の列を
列挙し、これをフィルタにかけて列の奇数だけが残るようにし、それぞれの要
素を⼆乗し、結果を合計します。
(define (sum-odd-squares tree)

(accumulate
+ 0 (map square (filter odd? (enumerate-tree tree)))))

even-fibsのほうでは、0から nまでの整数を列挙し、これらの整数のそれぞ
れに対するフィボナッチ数を⽣成し、結果の列をフィルタにかけて偶数の要素
だけが残るようにし、結果をリストに集積します。
(define (even-fibs n)

(accumulate
cons

14これは、実際は練習問題 2.28の fringe⼿続きそのものです。ここでは、この⼿続き
が汎⽤的な列操作⼿続きに属するものであることを強調するために名前をつけ替えてい
ます。

nil
(filter even? (map fib (enumerate-interval 0 n)))))

プログラムを列の演算として表すことの利点は、モジュール化された形でのプ
ログラムの設計がやりやすくなるということにあります。モジュール化とは、
⽐較的独⽴した部品を組み⽴ててプログラムを構築する設計のことです。標準
コンポーネントのライブラリとともに、コンポーネントを柔軟に接続できる標
準インターフェイスを提供することで、モジュール化された設計を促すことが
できます。
モジュールによる設計は、⼯学の設計において複雑性をコントロールする

強⼒な戦略です。例えば、実際の信号処理の応⽤では、標準化されたフィルタ
や変換器から要素を選び、それを直列につなぐことによってシステムを構築す
るということを、設計者は⽇常的に⾏っています。同じように、列の演算は⾃
由に組み合わせられる標準的なプログラムの要素のライブラリを提供します。
例として、最初の n+ 1個のフィボナッチ数の⼆乗のリストを構築するプログ
ラムを作るのに、sum-odd-squares⼿続きと even-fibs ⼿続きの部品を再利
⽤できます。
(define (list-fib-squares n)

(accumulate
cons
nil
(map square (map fib (enumerate-interval 0 n)))))

(list-fib-squares 10)
(0 1 1 4 9 25 64 169 441 1156 3025)

部品を配置し直して、整数列の中の奇数のものの⼆乗の積を計算するために使
うこともできます。
(define (product-of-squares-of-odd-elements sequence)

(accumulate * 1 (map square (filter odd? sequence))))
(product-of-squares-of-odd-elements (list 1 2 3 4 5))
225

⼀般的なデータ処理アプリケーションを列の演算として定式化することもで
きます。⼈事記録の列があるとして、最も給料の⾼いプログラマの給料を⾒つ
けたいとします。ある⼈事記録に含まれる給料を返す salaryというセレクタ
と、ある⼈事記録がプログラマのものであるかをチェックする programmer?と

いうセレクタが⽤意されているとします。すると、次のように書くことができ
ます。
(define (salary-of-highest-paid-programmer records)

(accumulate max 0 (map salary (filter programmer? records))))

これらの例は、列の演算として表すことができる広範囲な演算のごく⼀部を垣
間⾒たにすぎません。15

ここでリストとして実装した列というものは、処理モジュールを接続でき
るようにする標準インターフェイスとして使うことができます。また、構造を
列として統⼀的に表現するときに、プログラム中でのデータ構造の依存関係が
少ない数の列の演算に局所化されるようにしています。これらを変えることで、
プログラムの全体的な設計に⼿を加えずに、列の表現⽅法をいろいろ試してみ
ることができます。3.5節で、無限列を扱えるように列処理のパラダイムを⼀
般化する際に、この能⼒を利⽤することになります。

練習問題 2.33: 基本的なリスト操作のいくつかを集積として定義
したものを以下に⽰す。⽋けている式を埋めて、完成させよ。
(define (map p sequence)

(accumulate (lambda (x y) ⟨??⟩) nil sequence))
(define (append seq1 seq2)

(accumulate cons ⟨??⟩ ⟨??⟩))
(define (length sequence)

(accumulate ⟨??⟩ 0 sequence))

練習問題 2.34: ある xの値によって xの多項式を評価することは
集積として定式化できる。次の多項式について考える。

anx
n + an−1x

n−1 + · · ·+ a1x+ a0

15Richard Waters (1979)は、伝統的な Fortranプログラムを⾃動的に解析し、プログ
ラムをマップ、フィルタ、集積の組み合わせとして⾒ることができるようにするプログ
ラムを開発しました。すると、Fortran科学サブルーチンパッケージの実に 90%が、こ
のパラダイムにうまくはまることがわかりました。Lispがプログラミング⾔語として成
功した理由のひとつには、順序つきのコレクションを表現する標準的な⼿段がリストと
して提供され、⾼階演算によって操作ができるということがあります。プログラミング
⾔語 APLも同じような選択をして、それが⼒と魅⼒の⼤きな源となっています。APL
では、すべてのデータは配列として表され、配列のさまざまな演算ができる普遍的で便
利な汎⽤演算⼦セットが提供されています。

この多項式をホーナー法 (Horner’s rule)というアルゴリズムによ
って評価する。ホーナー法では、この計算を以下のような構造に
する。

(. . . (anx+ an−1)x+ · · ·+ a1)x+ a0.

つまり、an から始めて、それに xをかけ、an−1 を⾜し、xをかけ
. . .ということを、a0 まで繰り返す。16

以下のひな形を埋め、多項式をホーナー法によって評価する⼿続
きを作れ。多項式の係数 a0 . . . an は列として並んでいるとする。
(define (horner-eval x coefficient-sequence)

(accumulate (lambda (this-coeff higher-terms) ⟨??⟩)
0
coefficient-sequence))

例えば、x = 2のときの 1 + 3x + 5x3 + x5 を計算するには、次を
評価する。
(horner-eval 2 (list 1 3 0 5 0 1))

練習問題 2.35: 2.2.2節の count-leavesを集積として再定義せよ。
(define (count-leaves t)

(accumulate ⟨??⟩ ⟨??⟩ (map ⟨??⟩ ⟨??⟩)))

練習問題 2.36:⼿続き accumulate-nは accumulateに似ているが、
三番⽬の引数として列の列を取る (要素となる列の⻑さは⼀定で
あるとする)という違いがある。この⼿続きは、指定された集積⼿
続きを適⽤してそれぞれの列の最初の要素を結合したもの、それ

16Knuth 1981によると、この⽅法は 19世紀初めにW. G. Hornerによって定式化さ
れたということですが、実際はそれより 100年以上前にニュートンによって使われてい
ました。ホーナー法のやり⽅では、直接的なやり⽅ (まず anxn を計算し、an−1xn−1 を
⾜し . . .)よりも少ない回数の和と積によって多項式を評価します。実際に、任意の多項
式を評価するどんなアルゴリズムも、少なくともホーナー法以上の和と積を使う必要が
あり、ホーナー法が多項式評価のための最適なアルゴリズムであるということが証明可
能です。このことは、(和の回数については)1954年の A. M. Ostrowskiの論⽂で証明さ
れ、それが実質的に最適アルゴリズムの現代的研究の始まりとなりました。積の回数に
ついての同じような証明は、1966年に V. Y. Panによってなされました。Borodin and
Munro (1975)の本では、これらを含めた最適アルゴリズムについて概観しています。

ぞれの列の⼆番⽬の要素を結合したもの . . .を返り値とする。例え
ば、sが ((1 2 3) (4 5 6) (7 8 9) (10 11 12))という四つの列
を持つ列であるときに (accumulate-n + 0 s)の値が (22 26 30)
になるようにする。次の accumulate-nの定義に⽋けている式を埋
めよ。
(define (accumulate-n op init seqs)

(if (null? (car seqs))
nil
(cons (accumulate op init ⟨??⟩)

(accumulate-n op init ⟨??⟩))))

練習問題 2.37: ベクトル v = (vi)を数値の列として表現し、⾏列
m = (mij)をベクトル (⾏列の⾏)の列として表現するとする。例
えば、以下の⾏列は  1 2 3 4

4 5 6 6

6 7 8 9


列 ((1 2 3 4) (4 5 6 6) (6 7 8 9))として表現される。この表
現を使うと、列操作によって基本的な⾏列とベクトルの演算を簡
潔に表現できる。これらの (どんな⾏列演算の本にも書いてある)
演算は、次のようなものである。

(dot-product v w) 総和Σiviwiを返す
(matrix-*-vector m v) ti = Σjmijvjであるような

ベクトルtを返す
(matrix-*-matrix m n) pij = Σkmiknkjであるような

⾏列pを返す
(transpose m) nij = mjiであるような

⾏列nを返す

内積は、次のように定義できる。17

(define (dot-product v w)
(accumulate + 0 (map * v w)))

17この定義では、Footnote 12で説明した拡張バージョンの mapを使っています。

ほかの⾏列演算を計算する以下の⼿続きについて、⽋けた式を補
え (⼿続き accumulate-nは練習問題 2.36で定義されている)。
(define (matrix-*-vector m v)

(map ⟨??⟩ m))
(define (transpose mat)

(accumulate-n ⟨??⟩ ⟨??⟩ mat))
(define (matrix-*-matrix m n)

(let ((cols (transpose n)))
(map ⟨??⟩ m)))

練習問題 2.38: accumulate ⼿続きは、列の最初の要素と、右の
すべての要素を組み合わせた結果とを組み合わせるため、fold-
rightとしても知られている。fold-leftというものもあり、これ
は fold-rightに似ているが、要素の組み合わせを逆⽅向に⾏うと
いう点が違う。
(define (fold-left op initial sequence)

(define (iter result rest)
(if (null? rest)

result
(iter (op result (car rest))

(cdr rest))))
(iter initial sequence))

以下の式の値はいくらか。
(fold-right / 1 (list 1 2 3))
(fold-left / 1 (list 1 2 3))
(fold-right list nil (list 1 2 3))
(fold-left list nil (list 1 2 3))

fold-right と fold-left が任意の列に対して同じ値を返すこと
を保証するために、opが満たさなければならない性質を答えよ。
練習問題 2.39:以下の reverse(練習問題 2.18)⼿続きの定義を、練
習問題 2.38の fold-rightと fold-leftによって完成させよ。
(define (reverse sequence)

(fold-right (lambda (x y) ⟨??⟩) nil sequence))

(define (reverse sequence)
(fold-left (lambda (x y) ⟨??⟩) nil sequence))

マップのネスト
列というパラダイムを拡張して、普通はネストしたループによって表現される
ような多くの計算に適⽤できます。18次の問題について考えてみましょう。“正
の整数 nが与えられたとき、1 ≤ j < i ≤ nで、かつ i+ j が素数となるような
異なる正の整数 iと j のすべての順序つきペアを⾒つけよ。” 例えば、nが 6
のとき、ペアは以下のようになります。

i 2 3 4 4 5 6 6
j 1 2 1 3 2 1 5

i+ j 3 5 5 7 7 7 11

この計算の⾃然な構築⽅法として、n以下の正の整数からなる、⼤きい順に並
んだすべてのペアの列を⽣成し、フィルタによって合計が素数となるペアを選
択し、それからフィルタを通過したそれぞれの (i, j)のペアに対して (i, j, i+ j)

という三つ組を作るというものがあります。
ペアの列を⽣成する⽅法としては、すべての整数 i ≤ nに対して整数 j < i

を列挙し、そのような iと jすべてに対して (i, j)を⽣成するというものがあり
ます。ここで、列の演算を使って、(enumerate-interval 1 n)という列にマッ
プを⾏います。そして、この列のそれぞれの iに対して、(enumerate-interval
1 (- i 1))という列にマップを⾏います。この後者の列のそれぞれの j に対
して、(list i j)というペアを⽣成します。これによって、それぞれの iに対
するペアの列ができます。すべての iに対して、すべての列を (appendで集積
して)組み合わせることで、求めるペアの列ができます。19

(accumulate
append nil (map (lambda (i)

(map (lambda (j) (list i j))

18このマップのネストというやり⽅は、David Turnerによって⽰されたものです。彼
が開発した⾔語の KRCとMirandaでは、これらの概念を扱うエレガントな形式が⽤意
されています。この節の例 (と練習問題 2.42)は、Turner 1981をアレンジしたものです。
3.5.3節では、このやり⽅が無限の⻑さの列に対して⼀般化できることを学びます。

19ここではペアを、Lispのペアとしてではなく、⼆つの要素のリストとして表現して
います。このため、“ペア”(i, j)は (list i j)で、(cons i j)ではありません。

(enumerate-interval 1 (- i 1))))
(enumerate-interval 1 n)))

マップと集積を appendによって組み合わせることは、この種のプログラムで
はとてもよくあることですので、独⽴した⼿続きとして分離することにします。
(define (flatmap proc seq)

(accumulate append nil (map proc seq)))

次に、このペアの列をフィルタして、和が素数となるものを探します。フィル
タの述語は、列のそれぞれの要素に対して呼ばれます。述語の引数はペアで、
述語はペアから整数を取り出す必要があります。そのため、列のそれぞれの要
素に適⽤する述語は次のようになります。
(define (prime-sum? pair)

(prime? (+ (car pair) (cadr pair))))

最後に、フィルタを通ったペアの列に対して次の⼿続きでマップして、結果の
列を⽣成します。⼿続きは、ペアの⼆つの要素とその合計からなる三つ組を構
築するものです。
(define (make-pair-sum pair)

(list (car pair) (cadr pair) (+ (car pair) (cadr pair))))

これらのステップすべてを組み合わせると、完全な⼿続きになります。
(define (prime-sum-pairs n)

(map make-pair-sum
(filter prime-sum? (flatmap

(lambda (i)
(map (lambda (j) (list i j))

(enumerate-interval 1 (- i 1))))
(enumerate-interval 1 n)))))

マップのネストは、区間の列挙以外の列に対しても役に⽴ちます。ある集合 S

に対して、そのすべての順列を⽣成したいとします。つまり、集合の項⽬を並
べる⽅法をすべて⽣成するということになります。例えば、{1, 2, 3}の順列は、
{1, 2, 3}, {1, 3, 2}, {2, 1, 3}, {2, 3, 1}, {3, 1, 2}, {3, 2, 1}です。集合 Sの順列を⽣
成するための計画として、S のすべての項⽬ xに対して、S − xの順列の列を
⽣成し、20 それぞれの先頭に xを置くというやり⽅が使えます。こうすると、

20集合 S − xは、S のすべての要素から xを除いた集合です。

S の xそれぞれに対して、xから始まる S の順列の列が得られます。すべての
xに対するこれらの列を組み合わせると、S の順列すべてが得られます。21

(define (permutations s)
(if (null? s) ; 集合は空か?

(list nil) ; 空集合を持つ列
(flatmap (lambda (x)

(map (lambda (p) (cons x p))
(permutations (remove x s))))

s)))

この戦略によって、S の順列を求める問題を、S よりも要素数の少ない集合の
順列を求める問題に縮約していることに注意してください。最終的には、要素
のない集合を表す空リストに到達します。この場合、(list nil)というひと
つの項⽬からなる列、つまり空集合を⽣成します。permutationsで使われる
remove⼿続きは、与えられた列から与えられた項⽬を除いたすべての項⽬を
返します。これは単純なフィルタとして表現できます。
(define (remove item sequence)

(filter (lambda (x) (not (= x item)))
sequence))

練習問題 2.40: 整数 nに対し、1 ≤ j < i ≤ nとなるペア (i, j)の
列を⽣成する⼿続き unique-pairsを定義せよ。unique-pairsを
使って上の prime-sum-pairsの定義を簡単にせよ。
練習問題 2.41: ある整数 n以下の異なる正の整数が⼤⼩順に並ん
だ三つ組 i, j, kの中で、合計がある整数 sとなるものすべてを⾒
つける⼿続きを書け。
練習問題 2.42:
“8クイーンパズル”とは、どのクイーンもほかのクイーンの利き
筋に⼊らない (つまり、同じ⾏・列・対⾓線の上に⼆つのクイーン
があるということがないようにする) ように 8個のクイーンをチ
ェス盤の上に置く⽅法を問うものである。図 2.8に考えられる解の

21Schemeのコードでは、セミコロンはコメント (comment) を⼊れる場合に使います。
セミコロンから⾏末までにあるものはすべてインタプリタに無視されます。この本では
あまりコメントを使っていません。プログラムに対して説明的な名前をつけることで、プ
ログラム⾃⾝がドキュメント化されたものとなるよう努⼒しています。

図 2.8: 8クイーンパズルのひとつの解

ひとつを⽰す。このパズルを解くひとつのやり⽅として、各列に
ひとつのクイーンを置きながらチェス盤を横に移動していくとい
うものがある。k − 1個のクイーンを置いた状態では、k個⽬のク
イーンは、すでに盤上にあるどのクイーンも利き筋に⼊らない位
置に置く必要がある。この解き⽅は再帰的に定式化できる。最初
の k− 1列に k− 1個のクイーンを置くパターンをすべて⽣成済み
だと想定する。それぞれのパターンに対して、k列⽬のそれぞれの
⾏にクイーンを置いていって、位置の集合を拡張したものを⽣成
する。次にこれらをフィルタして、k列⽬のクイーンがほかのクイ
ーンに利かないものだけを残す。こうすると、最初の k列に k個
のクイーンを置くすべてのパターンの列ができる。このプロセス
を続けると、パズルの解答のひとつだけではなく、すべての解答
が得られる。
この解き⽅を queensという⼿続きとして実装し、n× nのチェス
盤に n個のクイーンを置くという問題に対するすべての答えの列
を返すようにする。queens は、盤の最初の k 列にクイーンを置
くすべてのパターンの列を返す内部⼿続き queen-colsを持って
いる。
(define (queens board-size)

(define (queen-cols k)

(if (= k 0)
(list empty-board)
(filter
(lambda (positions) (safe? k positions))
(flatmap
(lambda (rest-of-queens)

(map (lambda (new-row)
(adjoin-position
new-row k rest-of-queens))

(enumerate-interval 1 board-size)))
(queen-cols (- k 1))))))

(queen-cols board-size))

この⼿続きの中で、rest-of-queensは最初の k−1列に k−1個の
クイーンを置くパターンのひとつで、new-rowは k列⽬のクイーン
を置く候補となる⾏である。盤上の位置集合に対する表現⽅法と、
位置集合に新しい⾏・列の位置を追加する adjoin-position⼿続
きと位置の空集合を表す empty-boardを実装し、プログラムを完
成させよ。また、位置集合に対して、k列⽬のクイーンがほかのク
イーンに利いていないかを調べる safe?⼿続きも書く必要がある
(新しいクイーンの利きだけをチェックすればいいということに注
意。ほかのクイーンは互いに利いていないことが保証済みである)。
練習問題 2.43: Louis Reasonerは練習問題 2.42を解くのにだいぶ
苦労している。彼が書いた queens ⼿続きは動いてはいるようだ
が、⾮常に遅いのだ (6× 6の場合でも、Louisは待ちきれなくなっ
てしまう)。Louisが Eva Lu Atorに⾒てくれるよう頼むと、彼⼥
は Louisが flatmap内のマップのネストの順番を次のように逆に
してしまっていることを指摘した。
(flatmap
(lambda (new-row)

(map (lambda (rest-of-queens)
(adjoin-position new-row k rest-of-queens))

(queen-cols (- k 1))))
(enumerate-interval 1 board-size))

逆にするとなぜプログラムの実⾏が遅くなるのか説明せよ。練習

問題 2.42のプログラムが 8クイーンパズルを解く時間を T として、
Louisのプログラムがパズルを解くのにかかる時間を⾒積もれ。

2.2.4 例: 図形⾔語
この節では、図形を描く簡単な⾔語を紹介し、それによって抽象と閉包の持つ
⼒を⽰し、また⾼階⼿続きの本質的な部分を利⽤します。この⾔語は、図 2.9の
ような、要素をずらしたり拡⼤縮⼩したりしたものによって構成されるパター
ンについての実験がやりやすいように設計されています。22 この⾔語では、組
み合わされるデータオブジェクトは、リスト構造ではなく⼿続きとして表現さ
れます。閉包性を満たす consによってどんな複雑なリスト構造でも簡単に構
築できるのと同じように、この⾔語の演算も閉包性を満たしていて、どんな複
雑なパターンでも簡単に構築できます。

図形⾔語
1.1節でプログラミングの学習を始めたとき、⾔語を記述する際には⾔語の基
本要素、組み合わせ⽅法、抽象化⽅法にフォーカスを当てることが重要だとい
うことを強調してきました。ここでもその枠組みに従います。
この図形⾔語のエレガントさの⼀部は、ペインタ (painter)という⼀種類の

要素しかないというところにあります。ペインタは、指定された平⾏四辺形型
の枠にフィットするように画像をずらしたり拡⼤縮⼩したりしたものを描画し
ます。例えば、waveという基本ペインタがあり、図 2.10のような粗い線画を描
きます。実際の線画の形は枠によって変わります—図 2.10の四つの画像は同じ
waveペインタによって⽣成されたものですが、それぞれ違った枠に対して描か
れています。もっと⼿の込んだペインタもあります。rogers という基本ペイン
タは、図 2.11に⽰すように、mit創設者の William Barton Rogersの絵を描き
ます。23 図 2.11の 4つの画像は、図 2.10の waveの画像と同じ 4つの枠に対し

22この図形⾔語は、M.C. Escherの “Square Limit” (Henderson 1982参照)という⽊
版画のような画像を構築するために Peter Hendersonが作成した⾔語をもとにしていま
す。この⽊版画は、この節の square-limit⼿続きによって描かれるものと似た、拡⼤縮
⼩した繰り返しパターンを含んでいます。

23William Barton Rogers (1804-1882) は mit の創設者で、また初代学⻑でもありま
す。地質学者であり、才能あふれる教師でもあった彼は、ウィリアム・アンド・メアリ
ー⼤学とバージニア⼤学で教鞭をとっていました。1859年にはボストンに移り、それま
でより研究の時間が持てるようになって、「⼯科⼤学」を設⽴する計画に取りかかりまし
た。また、マサチューセッツ州で初めてのガスメータの州検査官も務めました。

図 2.9: 図形⾔語によって⽣成されたデザイン

1861年に mitが創設されたとき、Rogersは最初の学⻑に選ばれました。Rogersは「役
に⽴つ学習」という理想を信奉していましたが、それは当時の⼤学教育とは異なるもの
でした。当時の⼤学教育は古典を過度に重視していて、彼の書くところによると「より幅
広く⾼度で実⽤的な⾃然・社会科学の教育や訓練を⾏ううえでの障害となっている」も
のでした。彼はまた、そこでの教育は職業専⾨学校の狭い教育とも異なるものにしよう
と思っていました。Rogerの⾔葉によると、

世間が押しつけてくるような、実務者と科学者の区別というのはまったく
不⽑なものだ。現代の経験はすべて、そのような区別が完全に無益だとい
うことを⽰している。

Rogersは 1870年に健康上の理由で辞任するまで mitの学⻑を務めました。1873年恐
慌による財政危機とハーバード⼤学による併合の試みをはねのけるという重圧から⼆代
⽬の mit学⻑ John Runkleが 1878年に辞職すると、Rogerは学⻑職に戻り、1881年ま
で務めました。

Rogersは 1882年の卒業式で mitの卒業⽣に演説をしているときに倒れ、亡くなりま
した。Runkleは、同年の弔辞の中で、Rogerの死に際の⾔葉を引⽤しています。

“本⽇ここに⽴ち、本校の現状を⾒ると . . . 科学の始まりのことが頭に浮か
びます。記憶によると、150年前に Stephen Halesが灯⽤ガスを主題にし
た⼩論⽂を発表しました。その中で彼は、彼の研究が⽰すところによると
128グレインの瀝⻘炭– ” “瀝⻘炭”というのが、この世界での彼の最後の

図 2.10: waveによって⽣成された画像

て描いたものです。
画像を結合するには、与えられたペインタから新しいペインタを構築する

いろいろな演算を使います。例えば、beside演算は⼆つのペインタを引数に取
り、枠の左半分に⼀つ⽬のペインタの画像を描き右半分に⼆つ⽬のペインタの
画像を描く新しい複合ペインタを作ります。同じように、belowは⼆つのペイ
ンタを引数に取り、⼆つ⽬のペインタの画像の下に⼀つ⽬のペインタの画像を
描く複合ペインタを作ります。単独のペインタを変形して別のペインタを作る

⾔葉でした。ここで彼は前のめりになり、まるで⽬の前の台に置いてある
メモを確認するかのようでしたが、それからゆっくりと直⽴の姿勢に戻り、
両⼿を上げ、そしてこの世の仕事と業績から “死の明⽇”へと送られたの
です。そこでは⼈⽣の謎は解決され、⾁体から解放された魂は、新しく未
だ計り知れない永遠の未来の謎について考え続けることに終わりのない満
⾜を得るのです。

Francis A. Walker(mitの三代⽬の学⻑)によると、
彼は⼈⽣を通して誠実かつ勇敢に⾏動し、騎⼠であれば必ず望んだような
よい死の迎え⽅をした。職務中に、持ち場について、公務を果たすちょう
どそのときに亡くなったのだ。

図 2.11: mit創設者・初代学⻑William Barton Rogersの画
像。図 2.10と同じ 4つの枠に対して描いたもの。(元の画像
は mit博物館の許諾を得て複製した)

演算もあります。例えば、flip-vertはひとつのペインタを引数に取り、その
画像を上下逆に描くプリンタを作ります。また、flip-horizは元のペインタ
の画像を左右逆に描くプリンタを作ります。

waveから始めて、⼆段階を経て構築した wave4というペインタの描く画像
を 図 2.12に⽰します。
(define wave2 (beside wave (flip-vert wave)))
(define wave4 (below wave2 wave2))

複雑な画像をこの⽅法で構築する際に、ペインタがこの⾔語の結合⼿段に
ついて閉じているということを利⽤しています。⼆つのペインタの besideや
belowは、それ⾃⾝ペインタになります。そのため、より複雑なペインタを作
るのに、それを要素として使うことができます。リスト構造を consを使って
構築する場合と同じように、データが結合⼿段について閉じているということ
は、少しの演算によって複雑な構造を作る能⼒を実現するうえで決定的に重要
なことです。
ペインタを結合できるとなると、よくある結合パターンを抽象化したくな

図 2.12: 図 2.10の waveペインタから始めて複雑な図を作る

るところです。ここで、ペインタ演算を Schemeの⼿続きとして実装すること
にします。つまり、この図形⾔語のための特別な抽象化メカニズムというもの
は必要ないということです。結合⼿段は普通の Scheme⼿続きなので、⼿続き
についてできることは何でもペインタ演算についてもできる能⼒が⾃動的に得
られることになります。例えば、wave4に出てくるパターンは次のように抽象
化できます。
(define (flipped-pairs painter)

(let ((painter2 (beside painter (flip-vert painter))))
(below painter2 painter2)))

そして、wave4はこのパターンの具体例として定義します。
(define wave4 (flipped-pairs wave))

また、再帰演算を定義することもできます。以下に⽰すものは、図 2.13と
図 2.14に⽰すような、右⽅向に枝分かれしていくペインタを作る再帰演算です。
(define (right-split painter n)

(if (= n 0)
painter
(let ((smaller (right-split painter (- n 1))))

(beside painter (below smaller smaller)))))

右⽅向だけでなく上⽅向にも枝分かれさせることで、バランスの取れたパター
ンを作ることもできます (練習問題 2.44と、図 2.13・ 図 2.14参照)。
(define (corner-split painter n)

(if (= n 0)
painter

right-split n

right-split

corner-split
up-
split

n--1

up-
split

right-split
identity

n--1n--1

n--1

n--1

corner-split n

n--1

n--1

図 2.13: right-splitと corner-splitの再帰計画

(let ((up (up-split painter (- n 1)))
(right (right-split painter (- n 1))))

(let ((top-left (beside up up))
(bottom-right (below right right))
(corner (corner-split painter (- n 1))))

(beside (below painter top-left)
(below bottom-right corner))))))

corner-splitの四つのコピーを適切に配置することによって、square-limit
というパターンを得ることができます。それを waveと rogersに適⽤したもの
を 図 2.9に⽰してあります。
(define (square-limit painter n)

(let ((quarter (corner-split painter n)))
(let ((half (beside (flip-horiz quarter) quarter)))

(below (flip-vert half) half))))

練習問題 2.44: corner-splitで使われている up-split⼿続きを
定義せよ。right-splitに似ているが、belowと besideの役割を

⼊れ替えている。

⾼階演算
ペインタを組み合わせるパターンを抽象化するだけでなく、より⾼いレベルで、
ペインタ演算を組み合わせるパターンを抽象化することもできます。つまり、
ペインタ演算を操作対象の要素として扱い、それらの要素を組み合わせる⼿段
—ペインタ演算を引数として取り、新しいペインタ演算を作るような⼿続き—
を書くことができるということです。
例えば、flipped-pairsと square-limitはどちらも、あるペインタの画像

を四つ正⽅形のパターンに並べるもので、違うのはそれぞれの画像の向きだけ
です。このペインタの組み合わせパターンを抽象化する⽅法のひとつを次の⼿
続きに⽰します。この⼿続きは、四つの⼀引数ペインタ演算を引数として取り、
ある与えられたペインタをこれらの四つの演算によって変形したものを正⽅形
に配置するペインタ演算を作るというものです。tl, tr, bl, brは、それぞれ左
上、右上、左下、右下の画像に適⽤する変換です。
(define (square-of-four tl tr bl br)

(lambda (painter)
(let ((top (beside (tl painter) (tr painter)))

(bottom (beside (bl painter) (br painter))))
(below bottom top))))

こうすると、flipped-pairsは square-of-fourによって以下のように定義で
きます。24

(define (flipped-pairs painter)
(let ((combine4 (square-of-four identity flip-vert

identity flip-vert)))
(combine4 painter)))

24等価な書き⽅として、次のように書くこともできます。
(define flipped-pairs

(square-of-four identity flip-vert identity flip-vert))

(right-split wave 4) (right-split rogers 4)

(corner-split wave 4) (corner-split rogers 4)

図 2.14:再帰演算 right-splitと corner-splitをペインタ
waveと rogersに適⽤したもの。corner-splitした四つの
図を組み合わせると、図 2.9に⽰すような対称的な square-
limitのデザインになる。

また、square-limitは次のように表現できます。25

(define (square-limit painter n)
(let ((combine4 (square-of-four flip-horiz identity

rotate180 flip-vert)))
(combine4 (corner-split painter n))))

練習問題 2.45: right-splitと up-splitは、汎⽤の分割演算の具
体例として表現できる。以下の式を評価したとき、
(define right-split (split beside below))
(define up-split (split below beside))

すでに定義したのものと同じふるまいを持つ⼿続き right-split
と up-splitが得られるように、split⼿続きを定義せよ。

枠
ペインタやペインタの組み合わせ⼿段をどのように実装するかを⽰す前に、ま
ず枠について考える必要があります。枠は三つのベクトルとして記述できま
す— 原点ベクトルと、⼆つの辺ベクトルです。原点ベクトルは、平⾯上のどこ
かにある絶対原点からの枠の原点の相対位置を指定します。⼆つの辺ベクトル
は、枠の原点を起点として、枠の⼆つの⾓の相対位置を指定します。もし⼆つ
の辺が垂直であれば、枠は⻑⽅形になります。そうでない場合、枠はより⼀般
的な平⾏四辺形になります。
図 2.15は、枠とそれに関連づけられたベクトルを⽰しています。データ抽

象化の原則通り、枠がどのように表現されているかについてここではまだ特定
する必要がありません。三つのベクトルを引数に取り枠を作るコンストラクタ
make-frameと、それに対応する三つのセレクタ origin-frame, edge1-frame,
edge2-frameがあるというだけで⼗分です (練習問題 2.47参照)。
画像は、単位正⽅形内の座標 (0 ≤ x, y ≤ 1)を使って記述することにしま

す。それぞれの枠には枠座標マップ (frame coordinate map) というものを関連
づけ、画像が枠に合うようにずらしたり拡⼤縮⼩したりするのに使います。こ

25rotate180はペインタを 180度回転するものです (練習問題 2.50参照)。rotate180の
代わりに、練習問題 1.42の compose⼿続きを使って (compose flip-vert flip-horiz)
と書くこともできます。

枠の辺1
ベクトル枠の辺2

ベクトル

枠の原点
ベクトル ディスプレイ上

での(0, 0)の点

図 2.15: 枠は、原点と⼆つの辺という三つのベクトルによっ
て記述される。

のマップは、ベクトル v = (x, y)を次のようなベクトルの和にマップすること
によって単位正⽅形を枠内に変換します。

Origin(Frame) + x · Edge1(Frame) + y · Edge2(Frame).

例えば、(0, 0)は枠の原点に、(1, 1)は原点と対⾓線上にある頂点に、(0.5, 0.5)
は枠の中⼼にマップされます。枠座標マップは次の⼿続きによって作ることが
できます。26

(define (frame-coord-map frame)
(lambda (v)

(add-vect
(origin-frame frame)
(add-vect (scale-vect (xcor-vect v) (edge1-frame frame))

(scale-vect (ycor-vect v) (edge2-frame frame))))))

26frame-coord-map は、後で練習問題 2.46で記述するベクトル演算を使っています。
これらは何らかのベクトルの表現を使って実装されているとします。データ抽象化のお
かげで、このベクトル表現がどのようなものであるかは、ふるまいさえ正しければ問題
になりません。

frame-coord-mapをある枠に適⽤すると、ベクトルが与えられるとベクトルを
返す⼿続きを返します。引数ベクトルが単位正⽅形内にあれば、返り値のベク
トルは枠内のものになります。例えば、次の式は
((frame-coord-map a-frame) (make-vect 0 0))

以下と同じベクトルを返します。
(origin-frame a-frame)

練習問題 2.46: 原点からある⼀点に張られる⼆次元ベクトルは、
x座標と y 座標からなるペアとして表現できる。コンストラクタ
make-vectと、それに対応するセレクタ xcor-vect, ycor-vectを
与え、ベクトルに対するデータ抽象化を実装せよ。それらのセレ
クタとコンストラクタによって、ベクトルの⾜し算、引き算、ス
カラによるかけ算という演算を⾏う⼿続き add-vect, sub-vect,
scale-vectを実装せよ。

(x1, y1) + (x2, y2) = (x1 + x2, y1 + y2),

(x1, y1)− (x2, y2) = (x1 − x2, y1 − y2),
s · (x, y) = (sx, sy).

練習問題 2.47: 次に、枠のコンストラクタとして使えるものを⼆
つ挙げる。
(define (make-frame origin edge1 edge2)

(list origin edge1 edge2))
(define (make-frame origin edge1 edge2)

(cons origin (cons edge1 edge2)))

それぞれのコンストラクタに対し、適切なセレクタを加えて枠の
実装を完成させよ。

ペインタ
ペインタは、枠を引数として取り、ある特定の画像をずらしたり拡⼤縮⼩した
りして枠にフィットするように描画する⼿続きとして表現できます。つまり、
pがペインタで fがフレームであれば、fを引数として pを呼び出すことで、f
の中に pの画像を⽣成します。

基本ペインタがどのように実装されるかという具体的なところは、グラフ
ィックシステムの特性と描画する画像のタイプに依存します。例えば、⼆つの
指定された点の間にスクリーン上の線を引く draw-lineという⼿続きを持っ
ているとします。すると、図 2.10の waveペインタのような線画のペインタは、
次のような線分のリストから作ることができます。27

(define (segments->painter segment-list)
(lambda (frame)

(for-each
(lambda (segment)

(draw-line
((frame-coord-map frame)
(start-segment segment))

((frame-coord-map frame)
(end-segment segment))))

segment-list)))

線分は、単位正⽅形に対する座標を使って与えられます。リスト中のそれぞれ
の線分に対し、ペインタは線分の両端点を枠座標マップを使って変換し、変換
した点の間に線を引きます。
ペインタを⼿続きとして表現することは、図形⾔語の中に強⼒な抽象化の

壁を作ります。さまざまなグラフィック機能に基づくいろいろな基本ペインタ
を作り、また組み合わせることができます。枠を引数として取り、その枠に合
わせて何かを描くものであれば、どんな⼿続きでもペインタとして使うことが
できます。

28

27segments->painter は、後で練習問題 2.48で記述する線分の表現を使っています。
また、練習問題 2.23で記述した for-each⼿続きも利⽤しています。

28例えば、図 2.11の rogersペインタはグレースケール画像によって構築されていま
す。枠内の各点に対し、rogersペインタは枠座標マップによってそこにマップされる画
像内の点を決定し、それに従って点の濃淡をつけます。2.1.3節では、有理数の実装はあ
る適切な条件を満たすものであれば何でもいいということを述べましたが、ここでさま
ざまな種類のペインタを使えるようにしているのは、そこで述べた抽象データという考
え⽅を利⽤したものです。ここでは、指定された枠の中に何かを描画するものであれば、
ペインタはどのような実装でもいいということを利⽤しています。2.1.3節ではペアを⼿
続きとして実装する⽅法を⽰しましたが、ペインタはデータを⼿続きによって表現する
⼆つ⽬の例となっています。

練習問題 2.48: 平⾯上の⽅向つきの線分は、ベクトルのペア—原
点から線分の始点へと張られるベクトルと、原点から線分の終点
へと張られるベクトル—として表現できる。練習問題 2.46で書い
たベクトル表現を使って、make-segmentというコンストラクタと
start-segmentと end-segmentというセレクタによって線分の表
現を定義せよ。
練習問題 2.49: segments->painterを使って、以下の基本ペイン
タを定義せよ。

a. 指定された枠の輪郭を描くペインタ。
b. 枠の対⾓線同⼠をつないで “X”を描くペインタ。
c. 枠の辺の中点をつないで菱形を描くペインタ。
d. waveペインタ。

ペインタの変形と組み合わせ
ペインタに対する演算 (flip-vertや besideのようなもの)は、引数として取
った枠から導出した枠に対して元のペインタを呼び出すようなペインタを作る
という仕組みで動いています。ですので、例えば flip-vertが画像をひっくり
返す際に、ペインタがどのように動作しているかを知る必要はありません—枠
をひっくり返す⽅法さえわかっていれば⼤丈夫です。ひっくり返したペインタ
は元のペインタをそのまま使いますが、それをひっくり返した枠に対して使う
というだけです。
ペインタ操作は、transform-painterという⼿続きに基づいています。こ

の⼿続きは引数として、ペインタと、枠をどのように変形するかという情報を
取り、新しいペインタを作ります。変形したペインタは、ある枠に対して呼ば
れたとき、その枠を変形して、変形した枠に対して元のペインタを呼び出しま
す。transform-painterの引数は (ベクトルとして表現した)複数の点で、新し
い枠の頂点を指定するものです。これらが枠に対してマップされるとき、⼀つ
⽬の点は新しい枠の原点を指定し、残り⼆つの点は枠の辺ベクトルの終点を指
定します。つまり、単位正⽅形内の引数は、元の枠内に含まれる枠を指定する
ことになります。
(define (transform-painter painter origin corner1 corner2)

(lambda (frame)
(let ((m (frame-coord-map frame)))

(let ((new-origin (m origin)))
(painter (make-frame

new-origin
(sub-vect (m corner1) new-origin)
(sub-vect (m corner2) new-origin)))))))

ペインタの画像を上下逆にするには次のようにします。
(define (flip-vert painter)

(transform-painter painter
(make-vect 0.0 1.0) ; new origin
(make-vect 1.0 1.0) ; new end of edge1
(make-vect 0.0 0.0))) ; new end of edge2

transform-painterを⽤いることで簡単に新しい変換を定義できます。右上 4
分の 1のフレームは次のようにして与えられます。
(define (shrink-to-upper-right painter)

(transform-painter
painter (make-vect 0.5 0.5)
(make-vect 1.0 0.5) (make-vect 0.5 1.0)))

画像を反時計回りに 90度回転させるという変形もあります。29

(define (rotate90 painter)
(transform-painter painter

(make-vect 1.0 0.0)
(make-vect 1.0 1.0)
(make-vect 0.0 0.0)))

また、画像を枠の中⼼⽅向につぶすという変形もあります。30

(define (squash-inwards painter)
(transform-painter painter

(make-vect 0.0 0.0)
(make-vect 0.65 0.35)
(make-vect 0.35 0.65)))

29rotate90は回転した枠にフィットするように画像を伸縮させるということも⾏うた
め、純粋な回転になるのは枠が正⽅形のときだけです。

30図 2.10と図 2.11に出てきた菱形の画像は、squash-inwardsを waveと rogersに適
⽤して作ったものです。

枠の変換は、⼆つ以上のペインタを組み合わせる⼿段を定義するためにキーと
なるものでもあります。例えば、beside⼿続きは⼆つのペインタを引数に取
り、それらを引数の枠の左半分と右半分を描画するように変形し、新しい複合
ペインタを作ります。ある枠を引数として複合ペインタを呼び出すと、⼀つ⽬
の変形ペインタを呼んで枠の左半分を描画し、⼆つ⽬の変形ペインタを呼んで
枠の右半分を描画します。
(define (beside painter1 painter2)

(let ((split-point (make-vect 0.5 0.0)))
(let ((paint-left

(transform-painter
painter1
(make-vect 0.0 0.0)
split-point
(make-vect 0.0 1.0)))

(paint-right
(transform-painter
painter2
split-point
(make-vect 1.0 0.0)
(make-vect 0.5 1.0))))

(lambda (frame)
(paint-left frame)
(paint-right frame)))))

ペインタのデータ抽象化と、特にペインタを⼿続きとして実装したことによっ
て、besideが実装しやすくなっているということがわかるでしょうか。部品と
なるペインタについて besideが知っておくべきことは、それぞれのペインタ
が指定した枠に何かを描画するということだけで、そのほかには何も知る必要
がありません。

練習問題 2.50: ペインタを左右逆にする変換 flip-horiz、ペイン
タを反時計回りに 180度、270度回転させる変換を定義せよ。
練習問題 2.51: ペインタに対する below演算を定義せよ。belowは
⼆つのペインタを引数に取る。返り値となるペインタは、枠が与
えられたとき、⼀つ⽬のペインタで枠の下部を描画し、⼆つ⽬のペ
インタで上部を描画する。⼆つのやり⽅で belowを定義せよ。⼀

つ⽬は、上に⽰した beside⼿続きと似たような⼿続きを書くとい
うもので、⼆つ⽬は、besideと適切な回転演算 (練習問題 2.50の
もの)によって定義するというものである。

頑健な設計のための⾔語レベル
図形⾔語は、⼿続きとデータによる抽象化についてこれまで紹介してきた重要
な考え⽅の練習となっています。基本的なデータ抽象化であるペインタは⼿続
きによる表現を使って実装され、異なる基礎的な描画機能を⾔語が統⼀したや
り⽅で扱うことができるようにしています。結合⼿段は閉包性を満たしている
ため、複雑な設計も簡単に組み⽴てることができます。これで、⼿続きを抽象
化するためのツールがすべて、ペインタの結合⼿段を抽象化するのに利⽤でき
るということになります。
また、⾔語とプログラム設計についての、もうひとつの決定的な考え⽅を

垣間⾒ることができました。それは階層化設計 (stratified design)というアプロ
ーチで、複雑なシステムは⼀連の⾔語によって記述される⼀連のレベルとして
構造化されるべきだという概念です。それぞれのレベルは、そのレベルで基本
とされる部品を組み合わせて構築され、それぞれのレベルで構築された部品は
次のレベルで基本部品として使われます。階層化設計のそれぞれのレベルで使
われる⾔語は、その詳細さのレベルに合った基本要素、結合⼿段、抽象化の⼿
段を備えています。
階層化設計は、複雑なシステムのエンジニアリングのあらゆるところに出

てきます。例えば計算機⼯学では、抵抗やトランジスタを組み合わせて (また
アナログ回路の⾔語を使って記述して)ANDゲートや ORゲートのような部品
を作り、それらはデジタル回路設計の⾔語では基本要素となります。31 これら
の部品を組み合わせてプロセッサ、バス構造、メモリシステムを構築し、今度
はコンピュータアーキテクチャに適した⾔語を使ってそれらを組み合わせてコ
ンピュータを作ります。コンピュータは、ネットワーク相互接続を記述するの
に適した⾔語を使って組み合わされて分散システムを構成します。こういった
繰り返しです。
階層化のごく⼩さな例として、ここでの図形⾔語は基本要素 (基本ペイン

タ)を使っていますが、それらの基本要素は、segments->painterの線分のリ
ストや rogersのようなペインタの濃淡の詳細点や線を規定する⾔語を使って
作られたものです。図形⾔語についての説明の⼤半は、これらの基本要素を組
み合わせることや、besideや belowのような幾何学的な結合器に焦点を当て

313.3.4節でそのような⾔語のひとつを記述します。

ていました。さらに、besideや belowを⾔語の基本要素と⾒なすことで、さら
に⾼いレベルでの作業も⾏いました。その⾔語の square-of-fourのような演
算は、幾何学的な結合器を組み合わせる際のよくあるパターンを捉えています。
階層化設計は、頑健 (robust)なプログラムを作りやすくします。頑健とい

うのは、仕様の⼩さな変更が、プログラム上でも相応に⼩さな変更ですむとい
うことです。例えば、図 2.9に⽰した、waveをもとにした画像を変更したいと
します。このとき、最も下のレベルに⼿を加え、waveの細かい⾒た⽬を変える
こともできます。また、中間レベルで、corner-splitが waveを複製するやり
⽅を変えることもできます。最も上のレベルで、square-limitが四隅の画像
を配置するやり⽅を変えることもできます。⼀般的に、階層化設計のそれぞれ
のレベルは、システムの特性を表現するうえでの異なる語彙を提供し、システ
ムを変更するための異なる能⼒を提供します。

練習問題 2.52: 上で述べた各レベルに⼿を加え、図 2.9に⽰した
waveによる square-limitを変更せよ。具体的には、次のようなも
のである。

a. 練習問題 2.49の wave基本ペインタに線分をいくつか加えよ
(例えば、笑顔を追加するなど)。

b. corner-split によって構築されるパターンを変更せよ (例
えば、up-splitや right-splitの画像を⼆つではなく⼀つ
だけ使うようにする)。

c. square-of-fourを使うほうの square-limitを修正し、四隅
を違うパターンで組み⽴てるようにせよ (例えば、正⽅形の
四隅で⼤きなロジャーズ⽒が外を向くようにするなど)。

2.3 記号データ
ここまで使ってきた複合データオブジェクトはすべて、究極的には数値から構
成されていました。この節では、任意の記号をデータとして扱う能⼒を導⼊し、
私たちの⾔語の表現能⼒を拡張します。

2.3.1 クォート
記号を作って複合データを作れるとしたら、次のようなリストが使えることに
なります。

(a b c d)
(23 45 17)
((Norah 12) (Molly 9) (Anna 7) (Lauren 6) (Charlotte 4))

記号を含むリストは、私たちの⾔語の次のような式と同じように⾒えるかもし
れません。
(* (+ 23 45)

(+ x 9))
(define (fact n)

(if (= n 1) 1 (* n (fact (- n 1)))))

記号を操作するためには、私たちの⾔語に新しい要素を⼊れる必要があります。
それは、データオブジェクトをクォート (quote)する能⼒です。(a b)という
リストを構築するという場合について考えてみましょう。これは、(list a b)
と書いてもうまくいきません。この式は、aや bといった記号そのものではな
く、その値 (value)を要素としたリストを作ってしまうからです。この問題は、
⾃然⾔語についてはよく知られています。⾃然⾔語では、単語や⽂が意味的な
実体となることもあり、⽂字の列 (統語的な実体)となることもあります。⾃然
⾔語では、単語や⽂をそのまま⽂字の列として扱うということを⽰すのに、引
⽤符を使うのが⼀般的です。例えば、“John”の最初の⽂字というと、それは当
然 “J” です。誰かに “あなたの名前を声に出して⾔ってください”と頼んだら、
その⼈の名前を⾔ってくれるでしょう。しかし、“ ‘あなたの名前’を声に出し
て⾔ってください”と頼んだら、“あなたの名前”という単語列を⾔ってもらえ
るはずです。ほかの⼈の発⾔を伝えるために、引⽤符をネストさせないといけ
なくなっていることに気をつけてください。32

32⾔語でクォートを使えるようにすると、等しいものは等しいもので置き換えられる
という概念を破壊するため、⾔語を単純に推論する能⼒を台無しにしてしまいます。例
えば、三は⼀⾜す⼆ですが、「三」という単語は「⼀⾜す⼆」という句ではありません。
クォートを使うと、ほかの式を操作する式を作れるようになるので、そういう意味でク
ォートは強⼒です (これについては、第 4章でインタプリタを書く際に⾒ることになりま
す)。しかし、ある⾔語の中で、その⾔語で書かれた⽂について表現する⽂というものを
許すと、「等しいもの同⼠は交換できる」ということが何を意味するのかについて、⼀貫
性のある原則を維持することがとても難しくなります。例えば、もし宵の明星は明けの
明星であると知っていれば、「宵の明星は⾦星である」という⽂から、「明けの明星は⾦
星である」と推論できます。しかし、「ジョンは宵の明星は⾦星であると知っている」と
いうことからは、「ジョンは明けの明星は⾦星であると知っている」を導くことはできま
せん。

リストや記号について、式として評価の対象にするのでなく、データオブ
ジェクトとして扱いたいというときにも、これと同じような慣例に従うことが
できます。しかし、ここでのクォートの形式は、⾃然⾔語の場合と違って、ク
ォートするオブジェクトの前に引⽤符 (伝統的にシングルクォートの記号 'を
使います)をひとつだけ置くというものです。Schemeの⽂法では、オブジェク
トを区切るのに空⽩や括弧を使っているため、クォートはひとつだけで問題あ
りません。このようなわけで、シングルクォートは次のオブジェクトをクォー
トするという意味になっています。33

これで、記号とその値を区別できるようになりました。
(define a 1)
(define b 2)
(list a b)
(1 2)
(list 'a 'b)
(a b)
(list 'a b)
(a 2)

リストを表⽰するときに使う慣習的な表現を使って複合オブジェクトを⼊⼒す
ることも、クォートによってできるようになります。34

(car '(a b c))
a

33表⽰する⽂字列を囲むのにはダブルクォートを使ってきましたが、シングルクォー
トはそれとは違うものです。シングルクォートはリストや記号を表すのに使われるのに
対して、ダブルクォートは⽂字列にしか使われません。この本では、⽂字列を使うのは
項⽬を表⽰する場合だけです。

34厳密に⾔うと、この引⽤符の使い⽅は、⾔語内のすべての複合式は括弧で囲まれてい
てリストのように⾒えるという⼀般的な規則を破っています。ここで、quoteという特
殊形式を導⼊することによって、この⼀貫性を取り戻すことができます。quoteの役割
はクォートと同じです。つまり、'aの代わりに (quote a)と⼊⼒したり、'(a b c)の代
わりに (quote (a b c))と⼊⼒したりできます。これはインタプリタの動作⽅式そのま
まです。引⽤符は、次に来る完全な式を quoteを使って包んで、(quote ⟨expression⟩)
を作るということを、⼀⽂字に省略したものにすぎません。これは重要なポイントです。
こうすることによって、インタプリタから⾒える式はすべてデータオブジェクトとして
操作できるという原則を維持できるからです。例えば、(car '(a b c))という式は (car
(quote (a b c)))と同じものなので、(list 'car (list 'quote '(a b c)))を評価
することによって構築できます。

(cdr '(a b c))
(b c)

この原則通りに、'()を評価すると空リストが得られます。これによって、変
数 nilは必要なくなります。
記号の操作に使われるもうひとつの基本要素は eq?です。これは⼆つの記

号を引数として取り、それらが同じであるかどうかをテストするものです。35

eq?を使うと、memqという便利な⼿続きを実装できます。これは記号とリスト
という⼆つの引数を取り、記号がリストに含まれていなければ (つまり、リス
トのどの項⽬とも eq?でなければ)、memq は偽を返します。そうでなければ、
そのリストの中で記号が最初に⾒つかったところから始まるサブリストを返し
ます。
(define (memq item x)

(cond ((null? x) false)
((eq? item (car x)) x)
(else (memq item (cdr x)))))

例えば、次の式の値は偽になります。
(memq 'apple '(pear banana prune))

⼀⽅、次の式の値は (apple pear)になります。
(memq 'apple '(x (apple sauce) y apple pear))

練習問題 2.53: 以下の式をそれぞれ評価したとき、それらに対し
てインタプリタは何を表⽰するか。
(list 'a 'b 'c)
(list (list 'george))
(cdr '((x1 x2) (y1 y2)))
(cadr '((x1 x2) (y1 y2)))
(pair? (car '(a short list)))
(memq 'red '((red shoes) (blue socks)))
(memq 'red '(red shoes blue socks))

35⼆つの記号が同じ順番の同じ⽂字で構成されているとき、それらは “同じ”であると
考えることができます。このような定義は、まだここでは取り組むことが難しい、プログ
ラミング⾔語における “同⼀性”の意味とは何かという深い問題を避けて通っています。
第 3章 (3.1.3節)では、この問題に戻ってくることになります。

練習問題 2.54: ⼆つのリストは、同じ要素を同じ順番で持つとき
に equal?であると⾔える。例えば、
(equal? '(this is a list) '(this is a list))

は真である。しかし、
(equal? '(this is a list) '(this (is a) list))

は偽である。より正確に⾔うと、記号が等しいかどうかという基
本的な eq?を使って、再帰的に equal?を定義できる。aと bは、
どちらも記号であってそれらが eq?であるか、どちらもリストで
あって、(car a)と (car b)が equal?であり、かつ (cdr a)と
(cdr b)が equal?である場合、equal?である。この考え⽅を使っ
て、equal?を⼿続きとして実装せよ。36

練習問題 2.55: Eva Lu Atorはインタプリタに次の式を⼊⼒した。
(car ''abracadabra)

予想外なことに、インタプリタは quoteという表⽰を返してきた。
このことを説明せよ。

2.3.2 例: 記号微分
記号操作の例として、またデータ抽象化の別の例として、代数式の記号微分を
⾏う⼿続きの設計について考えてみましょう。⼿続きは、引数として代数式と
変数を取り、その変数に関するその式の導関数を返すようにします。例えば、
⼿続きに対する引数が ax2 + bx+ cと xであれば、⼿続きは 2ax+ bを返すと
いうものです。記号微分は、Lispにとっては特別な歴史的意義があります。記
号微分は、記号を操作するコンピュータ⾔語を開発するうえでの動機となる実
例のひとつでした。さらに、記号微分の研究は、現在応⽤数学者や物理学者に
使われるようになってきている強⼒な記号数式処理システムの開発につながる
⼀連の研究のスタート地点となるものです。

36実際は、プログラマは記号だけでなく数値も含むリストに対して equal?を使うこと
もあります。数値は記号とは⾒なされません。数として等しい⼆つの数値 (=でテストで
きます)が eq?でもあるかという問題は、⾼度に実装依存です。equal?の定義としては、
(Schemeで基本要素となっているもののように)aと bが両⽅とも数値である場合、aと
bが数として等しければそれらは equal?であるということも規定しておくほうがよいで
しょう。

記号微分プログラムを開発するにあたっては、2.1.1節の有理数システムを
開発したときと同じデータ抽象化戦略に従います。つまり、まず “和”、“積”、
“変数”といった抽象オブジェクトの上で動く微分アルゴリズムを定義するこ
とにして、それらの抽象オブジェクトがどのように表現されるべきかというこ
とは考えないことにします。表現の問題については後ほど取り組みます。

抽象データによる微分プログラム
問題を簡単にするために、⼆引数の⾜し算・かけ算という演算だけから構成さ
れる式だけを扱う、とても単純な記号微分プログラムについて考えることにし
ます。任意のそのような式の微分は、以下の簡約規則を適⽤することによって
⾏うことができます。

dc
dx = 0, c が定数またはx と異なる変数のとき,

dx
dx = 1,

d (u + v )
dx =

du
dx +

dv
dx ,

d (uv )
dx = u

dv
dx + v

du
dx .

後ろの⼆つの規則は本質的に再帰的であるということがわかるでしょうか。と
いうのは、和の導関数を求める際には、まずそれぞれの項の導関数を求め、そ
れらを⾜すことになるからです。それぞれの項は、さらに分解が必要な式であ
るかもしれません。項をどんどん⼩さなものに分解していくと、最終的には定
数か変数のものになり、その導関数は 0または 1になります。
これらの規則を⼿続きという形にするために、有理数の実装を設計したと

きと同じように、ちょっとした希望的思考をすることにします。もし代数式の
表現⼿段があるとするなら、ある式が和なのか、積なのか、定数なのか、変数
なのか識別できるはずです。また、式から構成要素を抽出することもできるで
しょう。和については、例えば加数 (第⼀項)と被加数 (第⼆項)を抽出できる
ようになっていてほしいところです。それに、構成要素から式を構築すること
もできるはずです。ここでは、次のようなセレクタ、コンストラクタ、述語を
実装する⼿続きがもうできているとします。
(variable? e) eは変数か?
(same-variable? v1 v2) v1と v2は同じ変数か?

(sum? e) eは和か?
(addend e) 和 eの加数
(augend e) 和 eの被加数
(make-sum a1 a2) a1と a2の和を構築する
(product? e) eは積か?
(multiplier e) 積 eの乗数
(multiplicand e) 積 eの被乗数
(make-product m1 m2) m1と m2の積を構築する
これらの⼿続きと、数値かどうかを判断する基本述語の number?を使って、微
分の規則を次の⼿続きのように表現できます。
(define (deriv exp var)

(cond ((number? exp) 0)
((variable? exp) (if (same-variable? exp var) 1 0))
((sum? exp) (make-sum (deriv (addend exp) var)

(deriv (augend exp) var)))
((product? exp)
(make-sum

(make-product (multiplier exp)
(deriv (multiplicand exp) var))

(make-product (deriv (multiplier exp) var)
(multiplicand exp))))

(else
(error "unknown expression type: DERIV" exp))))

この deriv⼿続きには、完全な微分アルゴリズムが組み込まれています。抽象
データによって表現されているため、ちゃんとしたセレクタとコンストラクタ
のセットを設計する限り、代数式をどう表現したとしても動くはずです。この
問題については次で⾒ていきます。

代数式を表現する
リスト構造を使って代数式を表現するやり⽅はいろいろ考えられます。例えば、
通常の代数記法を反映した記号のリストを使って、ax+ bを (a * x + b) とい
うリストとして表現するということも可能です。ですが、とりわけ素直なやり
⽅は、Lisp が複合式に使うのと同じ、括弧でくくった前置記法を使うというも

のです。つまり、ax+ bは (+ (* a x) b)と表現することになります。そうす
ると、微分問題に対するデータ表現は次のようになります。

• 変数は記号である。基本述語 symbol?で識別する。
(define (variable? x) (symbol? x))

• ⼆つの変数は、それらを表現する記号が eq?であれば等しい。
(define (same-variable? v1 v2)

(and (variable? v1) (variable? v2) (eq? v1 v2)))

• 和と積は、リストとして構築する。
(define (make-sum a1 a2) (list '+ a1 a2))
(define (make-product m1 m2) (list '* m1 m2))

• 和は、最初の要素が記号 +であるリストである。
(define (sum? x) (and (pair? x) (eq? (car x) '+)))

• 加数は、和のリストの⼆つ⽬の項である。
(define (addend s) (cadr s))

• 被加数は、和のリストの三つ⽬の項である。
(define (augend s) (caddr s))

• 積は、最初の要素が記号 *であるリストである。
(define (product? x) (and (pair? x) (eq? (car x) '*)))

• 乗数は、積のリストの⼆つ⽬の項である。
(define (multiplier p) (cadr p))

• 被乗数は、積のリストの三つ⽬の項である。
(define (multiplicand p) (caddr p))

こうすると、derivに具体的に書いたようなアルゴリズムによってこれらの要
素を組み合わせるだけで、ちゃんと動く記号微分プログラムができます。その
ふるまいについて、いくつかの例を⾒てみましょう。

159



(deriv '(+ x 3) 'x)
(+ 1 0)
(deriv '(* x y) 'x)
(+ (* x 0) (* 1 y))
(deriv '(* (* x y) (+ x 3)) 'x)
(+ (* (* x y) (+ 1 0))

(* (+ (* x 0) (* 1 y))
(+ x 3)))

プログラムは正しい答えを返します。しかし、答えは簡約されていません。次
の結果は、確かに正しいものです。

d (xy )
dx = x · 0 + 1 · y

しかし、プログラムには x · 0 = 0, 1 · y = y, 0 + y = yだということをわかって
いてほしいものです。⼆つ⽬の例の答えは、単純に yであるべきです。三つ⽬
の例でわかるように、式が複雑であるとこれは深刻な問題になってきます。
ここで問題になっていることは、有理数の実装の際に問題になったことと

よく似ています。答えが最も単純な形に簡約されていないという問題です。有
理数の簡約は、実装の中でコンストラクタとセレクタだけを変更するだけでで
きました。ここでも、同じような戦略が使えます。derivにはまったく⼿を加
えないことにします。その代わり、make-sumに変更を加え、もし両⽅の加数が
数値であれば、それらを⾜し合わせて和を返すようにします。また、加数のひ
とつが 0であれば、もうひとつの加数のみを返すようにします。
(define (make-sum a1 a2)

(cond ((=number? a1 0) a2)
((=number? a2 0) a1)
((and (number? a1) (number? a2))
(+ a1 a2))

(else (list '+ a1 a2))))

ここでは、ある式が与えられた数値と等しいかどうかをチェックする =number?
という⼿続きを使っています。
(define (=number? exp num) (and (number? exp) (= exp num)))

同じように、make-productに変更を加えて、0には何をかけても 0で、1に何
かをかけるとその何か⾃⾝になるという規則を組み込みます。

(define (make-product m1 m2)
(cond ((or (=number? m1 0) (=number? m2 0)) 0)

((=number? m1 1) m2)
((=number? m2 1) m1)
((and (number? m1) (number? m2)) (* m1 m2))
(else (list '* m1 m2))))

今回のものが、先ほどの三つの例に対してどう動作するかを次に⽰します。
(deriv '(+ x 3) 'x)
1
(deriv '(* x y) 'x)
y
(deriv '(* (* x y) (+ x 3)) 'x)
(+ (* x y) (* y (+ x 3)))

これでだいぶよくなりましたが、三つ⽬の例を⾒ると、式を誰もが “最も単純”
だといえるような形にするプログラムができるまでには、まだだいぶ距離があ
りそうです。代数的簡約化の問題は、何よりも、ある⽬的のためには最も単純
である形式がほかの⽬的のためにはそうではないということがありうるため、
複雑なものになっています。

練習問題 2.56: より多くの種類の式を扱えるようにこの基本的な
微分プログラムを拡張するにはどうすればよいかを⽰せ。例えば、
次の微分規則を実装せよ。

d (un )

dx = nun−1 du
dx

実装にあたっては、derivプログラムに新しい節を追加し、また
exponentiation?, base, exponent, make-exponentiationを適切
に定義せよ (冪乗の表現には **という記号を使ってもよい)。任意
の数の 0乗は 1であり、任意の数の 1乗はそれ⾃⾝であるという
規則を組み込め。
練習問題 2.57: 微分プログラムを拡張し、(2以上の)任意の数の項
の和と積を扱えるようにせよ。すると上の最後の例は以下のよう
に表現できる。
(deriv '(* x y (+ x 3)) 'x)

deriv⼿続きにはまったく⼿を加えず、和と積の表現のみを変更
することによって解け。例えば、和の addend(加数)は最初の項で、
augend(被加数)は残りの項の和というようにする。
練習問題 2.58: 微分プログラムを修正して、+と *が前置演算⼦で
はなく中置演算⼦となるような、通常の数学の記法に対して動作
させたいとする。微分プログラムは抽象データによって定義され
ているので、微分プログラムの動作基盤となる代数式の表現を定
義する述語、セレクタ、コンストラクタを変更するだけで、異なる
表現を扱うように修正できる。

a. (x + (3 * (x + (y + 2))))のような中置記法で表された代
数式を微分するにはどのようにするかを⽰せ。問題を簡単に
するため、+と *は常に⼆つの引数を取り、式は完全に括弧
でくくられていると仮定せよ。

b. (x + 3 * (x + y + 2))のような標準的な代数記法を認め
ると、問題はずっと難しくなる。この記法では、必要のない
括弧は省略し、乗算は加算より先に⾏われると仮定している。
ここでの微分プログラムがそのような記法に対してもうまく
動くように、適切な述語、セレクタ、コンストラクタを設計
できるだろうか。

2.3.3 例: 集合を表現する
ここまでの例では、有理数と代数式という、⼆種類の複合データオブジェクト
の表現を構築しました。この例のうちのひとつでは、式の単純化 (簡約化)をコ
ンストラクタで⾏うかセレクタで⾏うかという選択肢がありましたが、そのほ
かの点では、これらの構造のリストによる表現の選択肢は素直なものでした。
しかし、集合の表現ということになると、表現をどう決めるかということはあ
まり明らかではありません。実際に、可能な表現は数多くあり、それぞれいく
つかの点でお互いに⼤きく異なっています。
ざっくりした⾔い⽅をすると、集合とはただの異なるオブジェクトの集ま

りです。より正確な定義をするには、データ抽象化の⼿法を使うことができま
す。つまり、集合に使われる演算を規定することによって “集合”を定義すると
いうことです。それらの演算は、union-set, intersection-set, element-of-
set?, adjoin-setです。element-of-set?は、与えられた要素が集合の構成要
素であるかどうかを判定する述語です。adjoin-setは、オブジェクトと集合

を引数に取り、元の集合の要素に加えて追加された要素を含む集合を返します。
union-setは、⼆つの集合の和集合、つまりどちらかの引数に含まれる要素を
すべて含む集合を計算します。intersection-setは、⼆つの集合の共通部分、
つまり引数のどちらにも含まれる要素のみを含む集合を計算します。データ抽
象という視点から⾒ると、上の解釈と整合性があるものでさえあれば、これら
の演算を実装するのにどのような表現でも⾃由に設計できます。37

順序なしリストとしての集合
集合を表現するやり⽅のひとつは、要素が⼆回以上現れることがないリストと
いうものです。空集合は空リストとして表現します。この表現では、element-
of-set?は 2.3.1節の memq⼿続きと似たものになります。ここでは eq?ではな
く equal?を使っているため、集合の要素は記号である必要はありません。
(define (element-of-set? x set)

(cond ((null? set) false)
((equal? x (car set)) true)
(else (element-of-set? x (cdr set)))))

これを使って、adjoin-setを書くことができます。追加するオブジェクトが
すでに集合の中にあるなら、単にその集合を返します。そうでなければ、cons
を使ってそのオブジェクトを集合を表すリストに追加します。
(define (adjoin-set x set)

(if (element-of-set? x set)
set
(cons x set)))

37さらに形式的に書くなら、“上の解釈と整合性がある”というところを、演算が次の
ような⼀連の規則を満たすというように規定できます。
任意の集合 Sと任意のオブジェクト xに対し、(element-of-set? x (adjoin-set x S))
は真である (ざっくり⾔うと、“集合にオブジェクトを追加すると、そのオブジェクトを
含む集合ができる”)。
任意の集合 S, Tと任意のオブジェクト xに対し、(element-of-set? x (union-set S
T))は (or (element-of-set? x S) (element-of-set? x T))に等しい。(ざっくり⾔
うと、“(union S T)の要素は、Sに含まれる要素か、Tに含まれる要素である”)。
任意のオブジェクト xに対し、(element-of-set? x '())は偽である (ざっくり⾔うと、
“空集合の要素となるオブジェクトは存在しない”)。

intersection-setには、再帰的な戦略が使えます。もし、set2と、set1の cdr
との共通部分を作ることができるなら、set1の carをそれに含めるかどうか
を決めるだけで⼗分です。しかし、これは (car set1)が set2にも含まれてい
るかどうかによって決まります。結果として、⼿続きは次のようになります。
(define (intersection-set set1 set2)

(cond ((or (null? set1) (null? set2)) '())
((element-of-set? (car set1) set2)
(cons (car set1) (intersection-set (cdr set1) set2)))

(else (intersection-set (cdr set1) set2))))

表現を設計する際に考慮するべき問題のひとつは効率です。この集合演算に必
要となるステップ数を考えてみましょう。どの演算も element-of-set?を使
っているので、この演算の速度は集合の実装全体の効率に⼤きな影響を与え
ます。ここで、あるオブジェクトが集合の要素かどうかチェックするために、
element-of-set?は集合全体を⾛査する必要があるかもしれません (最悪の場
合、オブジェクトは集合の中になかったという結果になります)。そのため、集
合が n個の要素を持っているとすると、element-of-set?には最⼤ nステッ
プかかるかもしれません。つまり、必要なステップ数は Θ(n)で増加するとい
うことです。adjoin-setはこの演算を使うので、やはり Θ(n)で増加します。
intersection-setは、set1の要素それぞれに対して element-of-set?のチェ
ックをするため、必要なステップ数は関係する集合の⼤きさの積に⽐例して増
加、つまり⼤きさ nの⼆つの集合に対して Θ(n2)で増加するということになり
ます。union-setについても同じことが⾔えます。

練習問題 2.59: 順序なしリストとして表現した集合に対する
union-set演算を実装せよ。
練習問題 2.60:上の例では、集合は重複のないリストとして表現す
るよう規定した。ここで、重複を許す場合について考えてみよう。
その場合、例えば {1, 2, 3}という集合は (2 3 2 1 3 2 2)というリ
ストとして表現することもできる。この表現に対して演算を⾏う
⼿続き element-of-set?, adjoin-set, union-set, intersection-
setを設計せよ。それぞれの効率は、重複なし表現に対する⼿続き
でそれに対応するものと⽐べてどうだろうか。重複なしの表現よ
りもこの表現のほうが向いているような応⽤はあるだろうか。

順序つきリストとしての集合
この集合演算を速くする⽅法のひとつとして、集合の要素が昇順に並ぶように
表現を変えるというものがあります。これをするためには、⼆つのオブジェク
トを⽐較してどちらが⼤きいか判断するための何らかの⽅法が必要です。例え
ば、記号を辞書順に⽐較したり、オブジェクトに⼀意の数値を割り振ることに
決めて対応する数値によって要素を⽐較したりといったことが考えられます。
ここでは話を簡単にするため、集合の要素が数値である場合についてのみ考え
ることにします。この場合、要素は >と <によって⽐較できます。数値の集合
を表現するのには、要素を昇順に並べたものを使います。上に挙げた最初の表
現では、{1, 3, 6, 10} という集合を表すのに要素をどのように並べてもかまいま
せんでしたが、新しい表現では (1 3 6 10)というリストしか許可しません。
順序に従って並べることのひとつの利点は element-of-set? で出てきま

す。項⽬があるかどうかを調べるのに、もう集合全体を⾛査する必要はありま
せん。探している項⽬よりも⼤きい要素に出会ったら、その項⽬が集合内にな
いということがわかります。
(define (element-of-set? x set)

(cond ((null? set) false)
((= x (car set)) true)
((< x (car set)) false)
(else (element-of-set? x (cdr set)))))

これはどれだけのステップ数の節約になるでしょうか。最悪の場合、探してい
る項⽬は集合内の⼀番⼤きなものかもしれず、その場合ステップ数は順序なし
表現の場合と同じです。しかし、いろいろな⼤きさの項⽬を検索するときのこ
とを考えると、場合によってはリストの始点のあたりで検索を終了できたり、
場合によってはやっぱりリストのほとんどを調べないといけなかったり、とな
ることが期待できます。平均的には、集合の約半分の項⽬を検査しなければい
けないということになるでしょう。このため、必要な平均ステップ数はだいた
い n/2になります。これは依然として Θ(n)という増加オーダーですが、それ
でもやはり、平均的には、前の実装に⽐べてステップ数を半分に節約できます。

intersection-set では、さらに⽬覚ましい⾼速化が達成できます。順序
なし表現では、この演算には Θ(n2)というステップ数が必要でした。それは、
set1の要素それぞれに対して set2の完全な⾛査を⾏っていたからでした。し
かし、順序つきの表現では、より賢い⽅法が使えます。まず、⼆つの集合の最
初の要素 x1と x2を⽐較します。もし x1と x2が等しければ、それは共通部分
の要素になり、共通部分の残りは⼆つの集合の cdrの共通部分ということにな

ります。では、x1が x2より⼩さい場合はどうなるか考えてみましょう。x2は
set2の最⼩要素なので、x1は set2のどこにも現れないと結論づけることがで
きます。つまり、共通部分にも含まれないということになります。このため共
通部分は、set2と、set1の cdrとの共通部分に等しくなります。同じように、
もし x2が x1よりも⼩さければ、求める共通部分は、set1と、set2の cdrと
の共通部分となります。⼿続きは次のようになります。
(define (intersection-set set1 set2)

(if (or (null? set1) (null? set2))
'()
(let ((x1 (car set1)) (x2 (car set2)))

(cond ((= x1 x2)
(cons x1 (intersection-set (cdr set1)

(cdr set2))))
((< x1 x2)
(intersection-set (cdr set1) set2))

((< x2 x1)
(intersection-set set1 (cdr set2)))))))

このプロセスで必要となるステップ数を⾒積もるには、各ステップで、共通部
分問題をより⼩さな集合の共通部分集合の計算に縮⼩していることに注意して
ください。縮⼩は、set1と set2のどちらかまたは両⽅から最初の要素を削除
することによって⾏われます。このため、必要なステップ数は、最⼤でも set1
と set2の⼤きさの合計で、順序なし表現のように⼤きさの積になることはあ
りません。これは、Θ(n2)というオーダーに対してΘ(n)というオーダーになる
ということで、そこそこの⼤きさの集合に対してもかなりの⾼速化になります。

練習問題 2.61: 順序つき表現を使った adjoin-set を実装せよ。
element-of-set?から類推して、順序つきであることの利点を⽣
かして、順序なしの表現に⽐べて平均的に半分のステップを必要
とする⼿続きを作るやり⽅を⽰せ。
練習問題 2.62: 順序つきリストとして表現された集合に対して、
union-setを Θ(n)で実装せよ。

⼆分⽊としての集合
集合の要素を⽊という形式で配置することで、順序つきリスト表現をさらに改
善できます。⽊の各ノードは集合の要素をひとつ持ち、それはノードの “エン

図 2.16: 集合 {1, 3, 5, 7, 9, 11}を表現するさまざまな⼆分⽊

トリ”と呼ばれます。また、ほかの⼆つのノード (ノードは空であることもあり
ます)へのリンクも持っています。“左”のリンクは、そのノードより⼩さな要
素を指していて、“右”のリンクはそのノードより⼤きな要素を指しています。
図 2.16に、集合 {1, 3, 5, 7, 9, 11}を表現するいくつかの⽊を⽰します。ある同じ
集合を⽊という形で表現するのには、数多くの⽅法がありえます。ここで、有
効な⽊の表現の条件として求めるものは、あるノードの左部分⽊のすべての要
素がそのノードの要素よりも⼩さく、右部分⽊のすべての要素がそれよりも⼤
きいということだけです。
⽊による表現の利点は次の通りです。ある数値 xがある集合に含まれてい

るかどうかをチェックしたいとします。まず、xとトップノードのエントリを
⽐較します。もし xがそれより⼩さければ、左部分⽊だけを探索すればいいと
いうことがわかります。もし xがそれより⼤きければ、右部分⽊だけを探索す
ればいいということです。ここで、もし⽊が “バランスが取れた”状態であれ
ば、各部分⽊の⼤きさは元の⽊の約半分です。そのため、⼀回のステップで、
⼤きさ nの⽊を探索するという問題を、⼤きさ n/2の⽊を探索するという問題
に縮⼩したことになります。⽊の⼤きさは各ステップで半分になるので、⼤き
さ nの⽊の探索に必要なステップ数は Θ(logn)で増加すると予測できます。38

⼤きな集合に対しては、これは以前の表現に⽐べてかなりの⾼速化になるでし
ょう。
⽊はリストを使って表現できます。各ノードは、ノードのエントリ、左部

38各ステップで問題の⼤きさが半分になるということは、対数増加の特徴的な性質で
す。このことについては、1.2.4節の⾼速冪乗アルゴリズムや、1.3.3節の区間⼆分探索法
⾒てきました。

分⽊、右部分⽊という三つの項⽬のリストになります。左または右の部分⽊が
空リストである場合は、そこにつながる部分⽊がないということを⽰します。
この表現は、次のような⼿続きで表すことができます。39

(define (entry tree) (car tree))
(define (left-branch tree) (cadr tree))
(define (right-branch tree) (caddr tree))
(define (make-tree entry left right)

(list entry left right))

これで、上で述べたような戦略を使って element-of-set?を書くことができる
ようになります。
(define (element-of-set? x set)

(cond ((null? set) false)
((= x (entry set)) true)
((< x (entry set))
(element-of-set? x (left-branch set)))

((> x (entry set))
(element-of-set? x (right-branch set)))))

集合に項⽬を追加するというのも同じように実装でき、必要ステップ数はやは
り Θ(logn)になります。項⽬ xを追加するには、xをノードのエントリと⽐較
し、右と左のどちらの枝に追加するべきかを決めます。そして、xを適切な枝
に追加したら、この新しく作った枝を元のエントリやほかの枝とくっつけ合わ
せます。もし xがエントリに等しければ、ノードをそのまま返します。もし空
の⽊に xを追加するよう求められたら、エントリが xで、右と左の枝が空であ
る⽊を⽣成します。この⼿続きを以下に⽰します。
(define (adjoin-set x set)

(cond ((null? set) (make-tree x '() '()))
((= x (entry set)) set)
((< x (entry set))
(make-tree (entry set)

(adjoin-set x (left-branch set))

39ここでは、集合は⽊によって表現し、⽊はリストによって表現しています。つまり、デ
ータ抽象化の上にデータ抽象化が構築されているということです。⼿続き entry, left-
branch, right-branch, make-treeは、リストによる表現といった特定のやり⽅から “⼆
分⽊”という抽象化を分離する⽅法と⾒なすことができます。

(right-branch set)))
((> x (entry set))
(make-tree (entry set) (left-branch set)

(adjoin-set x (right-branch set))))))

⽊の探索が対数的ステップで実⾏できるという上の主張は、⽊が “バランスが
取れている” 訳注 4 という前提に依存しています。つまり、すべての⽊の左右
の部分⽊はだいたい同じ数の要素を持っていて、各部分⽊は親の約半分の要素
を持っているということです。しかし、ここで構築する⽊がバランスが取れ
ていると当てにできるでしょうか。バランスの取れた⽊から始めたとしても、
adjoin-setで要素を追加すると、バランスの取れていない結果になってしま
うかもしれません。新たに追加する要素の位置は、その要素と集合にすでに存
在する項⽬の⼤⼩を⽐べた結果によるので、要素を “ランダムに”追加すれば、
平均的には⽊はバランスの取れたものになりやすくなります。しかし、このこ
とは保証されていません。例えば、もし空集合から始めて、1から 7の数値を
順番に追加していくと、図 2.17に⽰すようなとてもバランスの悪い⽊ができあ
がってしまいます。この⽊では、すべての左部分⽊は空なので、単純な順序つ
きリストに対する優位点がありません。この問題を解決する⽅法のひとつとし
て、任意の⽊を同じ要素を持つバランスの取れた⽊に変換する演算を定義する
というものがあります。こうすると、何回か adjoin-set演算を⾏うたびにこ
の変換を実⾏して、集合をバランスの取れた状態にしておくことができます。
この問題を解く⽅法はほかにもありますが、だいたいは検索や挿⼊がどちらも
Θ(logn)ステップで⾏えるような新しいデータ構造を設計するということをし
ています。40

練習問題 2.63: 次の⼆つの⼿続きは、どちらも⼆分⽊をリストに
変換する。
(define (tree->list-1 tree)

(if (null? tree)
'()
(append (tree->list-1 (left-branch tree))

(cons (entry tree)

訳注 4「バランスの取れた⽊」に対しては “平衡⽊”、“バランス⽊”などの訳語があります
が、ここではバランスの取れた (balanced)との兼ね合いから「バランスの取れた⽊」と
しています。

40このようなデータ構造の例としては、B⽊ (B-tree)や ⾚黒⽊ (red-black tree)など
があります。この問題については多くの⽂献があります。Cormen et al. 1990参照。

図 2.17: 1から 7まで順番に追加するとできるバランスの取
れていない⽊

(tree->list-1
(right-branch tree))))))

(define (tree->list-2 tree)
(define (copy-to-list tree result-list)

(if (null? tree)
result-list
(copy-to-list (left-branch tree)

(cons (entry tree)
(copy-to-list

(right-branch tree)
result-list)))))

(copy-to-list tree '()))

a. ⼆つの⼿続きは、すべての⽊に対して同じ結果を⽣成するだ
ろうか。そうでないとしたら、結果はどのように異なるだろ
うか。⼆つの⼿続きは、図 2.16の⽊に対してどのようなリス
トを⽣成するだろうか。

b. ⼆つの⼿続きは、n要素のバランスの取れた⽊を変換する際
に、同じステップ数の増加オーダーを持っているだろうか。
そうでないなら、どちらが遅いだろうか。

練習問題 2.64: 次の⼿続き list->treeは、順序つきリストをバラ
ンスの取れた⽊に変換する。補助⼿続き partial-treeは、整数 n

と、少なくとも n個の要素を持つリストを引数に取り、リストの最
初の n個の要素を含むバランスの取れた⽊を⽣成する。partial-
treeの返り値はペア (consで構築される)で、carには構築された
⽊を持ち、cdrには⽊に含まれなかった要素のリストを持つ。
(define (list->tree elements)

(car (partial-tree elements (length elements))))
(define (partial-tree elts n)

(if (= n 0)
(cons '() elts)
(let ((left-size (quotient (- n 1) 2)))

(let ((left-result
(partial-tree elts left-size)))

(let ((left-tree (car left-result))
(non-left-elts (cdr left-result))
(right-size (- n (+ left-size 1))))

(let ((this-entry (car non-left-elts))
(right-result
(partial-tree
(cdr non-left-elts)
right-size)))

(let ((right-tree (car right-result))
(remaining-elts
(cdr right-result)))

(cons (make-tree this-entry
left-tree
right-tree)

remaining-elts))))))))

a. partial-treeがどのように動くのか、できるだけ明確に⽂
章で答えよ。リスト (1 3 5 7 9 11)に対して list->treeが
⽣成する⽊を描け。

b. list->treeが n要素のリストを変換するのに必要なステッ
プ数の増加オーダーはどのようになるだろうか。

練習問題 2.65: 練習問題 2.63と練習問題 2.64の結果を使って、(バ
ランスの取れた)⼆分⽊として実装された集合に対する union-set
と intersection-setを Θ(n)で実装せよ。41

集合と情報検索
ここまで、集合をリストで表現するのにどのような⽅法があるかを検討し、デ
ータオブジェクトに対してどのような表現を選ぶかということが、そのデータ
を使うプログラムのパフォーマンスに⼤きな影響を及ぼすということを⾒てき
ました。集合に対して詳しく検討するもうひとつの理由は、ここで考察するテ
クニックは情報検索に関わる応⽤の中でよく出てくるものだということです。
個別のレコードを⼤量に含むデータベースについて考えてみましょう。例

えば、企業の⼈事ファイルや、会計システムの取引などです。データ管理シス
テムは普通、そのレコードにアクセスしたり更新したりするのに多くの時間を
消費するので、レコードにアクセスする効率的な⼿法が必要となります。これ
は、各レコードの中で、識別キー (key)として使える部分を特定することによ
って⾏うことができます。キーは、レコードを⼀意に識別するものであれば何
でもかまいません。⼈事ファイルであれば、従業員番号などがよいでしょう。
会計システムであれば、取引番号になるかもしれません。キーが何であっても、
レコードをデータ構造として定義する際には、あるレコードに対して関連づけ
られたキーを取得する keyセレクタ⼿続きを持つ必要があります。
ここで、データベースをレコードの集合として表現することにします。ある

キーに対するレコードの場所を調べるには、lookupという⼿続きを使います。
この⼿続きは、キーとデータベースを引数に取り、そのキーを持つレコードを
返すか、そのようなレコードがなければ偽を返します。lookupは、element-
of-set?とほぼ同じように実装されます。例えば、レコードの集合が順序なし
リストとして表現されていれば、次のようにできます。
(define (lookup given-key set-of-records)

(cond ((null? set-of-records) false)
((equal? given-key (key (car set-of-records)))
(car set-of-records))

(else (lookup given-key (cdr set-of-records)))))

もちろん、巨⼤な集合を表現するのには、順序なしリストよりもいい⽅法があ
ります。レコードを “ランダムアクセス”する必要がある情報検索システムで

41練習問題 2.63から練習問題 2.65までは Paul Hilfingerによるものである。

は、普通は⽊構造による⼿法によって表現されます。例えば、前に検討した⼆
分⽊のようなものです。そのようなシステムを設計するにあたっては、データ
抽象化の⽅法論はとても役に⽴ちます。設計する⼈は、順序なしリストのよう
な単純で素直な表現を使って、最初の実装を作ることができます。これは最終
的なシステムのためには不適切なものでしょうが、システムの残りの部分をテ
ストするための “quick and dirty”(⼿早く汚い)データベースを提供してくれる
という意味では役に⽴ちます。データ表現を修正してより洗練されたものにす
ることは後になってからでもできます。もしデータベースが抽象セレクタとコ
ンストラクタによってアクセスされるようになっていれば、そうやって表現を
変更しても、システムの残りの部分には修正の必要がないはずです。

練習問題 2.66:
レコードの集合が、キーの数値の⼤⼩によって順序づけられた⼆
分⽊という構造になっている場合について、lookup⼿続きを実装
せよ。

2.3.4 例: ハフマン符号化⽊
この節では、集合と⽊を操作するためのリスト構造とデータ抽象化の使い⽅を
練習します。ここでは、データを 1と 0の列 (ビット列)として表現する⼿法を
適⽤対象とします。例えば、コンピュータ内でテキストを表現するのに使われ
る ascii標準符号は、各⽂字を 7ビットの列として表現します。7ビットを使
うことで、27、つまり 128通りの異なる⽂字を区別できます。⼀般的に、n個
の異なる記号を区別しようと思うと、記号ごとに log2nビットを使う必要があ
ります。もしメッセージすべてが A, B, C, D, E, F, G, Hという 8個の記号か
らなるとすると、⼀⽂字あたり 3ビットの符号を選ぶことができます。例えば、
次のようになります。

A 000 C 010 E 100 G 110
B 001 D 011 F 101 H 111

この符号を使って次のメッセージ

BACADAEAFABBAAAGAH

を符号化すると、54ビットの列になります。

001000010000011000100000101000001001000000000110000111

ascii符号や、上に挙げた Aから Hまでの符号は、メッセージ内のそれぞれの
記号を同じビット数で表すので、固定⻑ (fixed-length)符号と⾔います。しか
し、時には異なる記号が異なるビット数で表されることもある可変⻑ (variable-
length) 符号を使ったほうがいい場合もあります。例えばモールス符号では、ア
ルファベットの各⽂字に対して使う短点と⻑点の数が異なります。特に、最も
よく出てくる⽂字の Eは、短点ひとつで表されます。⼀般的に、もしメッセー
ジの中でとてもよく出てくる記号とほとんど出てこない記号があるのなら、よ
く出てくる記号に短い符号を割り当てるとデータをより効率的に (つまり、メ
ッセージに対して少ないビット数で)符号化できます。次に挙げる、Aから H
までに対する別のコードについて考えてみましょう。

A 0 C 1010 E 1100 G 1110
B 100 D 1011 F 1101 H 1111

この符号を使って符号化すると、上と同じメッセージが次のようになります。

100010100101101100011010100100000111001111

この列は 42ビットなので、上に⽰した固定⻑符号に⽐べて空間を 20%以上節
約できていることになります。
可変⻑符号を使ううえで難しいことのひとつに、0と 1の列を読んでいく

際に、いつ記号の終わりに達したかをどうやって知るかということがあります。
モールス符号では、各⽂字に対応する短点と⻑点の列の後に、特別な分離符号
(separator code)(この場合は休⽌)を置くことで、この問題を解決しています。
ほかの解決法として、どの記号についても、その符号全体がほかの記号に対す
る符号の前の部分 (接頭辞 (prefix))になることがないように符号を設計すると
いうものがあります。そのような符号は接頭符号 (prefix code)と呼ばれます。
上の例では、Aを 0、Bを 100として符号化しているので、ほかのどの記号も、
0や 100で始まる符号を持つことはできません。
⼀般的に、符号化するメッセージの相対頻度を利⽤した可変⻑接頭符号を

使えば、かなりの節約ができます。これを⾏う戦略のひとつにハフマン符号化
法というものがあります。これは発⾒者の David Huffmanにちなんでつけら
れた名前です。ハフマン符号は⼆分⽊として表現でき、葉は符号化対象の記号
となります。⽊の中で、葉でないノードは、そのノード以下の葉に対応する記
号をすべて含んだ集合を持っています。それに加えて、葉の記号それぞれには
重み (相対頻度)が割り当てられていて、葉でないノードはその下にある葉の重
みをすべて合計した重みをそれぞれ持っています。重みは符号化・復号化の際

{A B C D E F G H} 17

{B C D E F G H} 9

A 8

{B C D} 5

{C D} 2

D 1C 1

B 3

{E F G H} 4

{G H} 2

{E F} 2

E 1 F 1

H 1G 1

図 2.18: ハフマン符号化⽊

には使いません。以下では、⽊を構築する際に重みをどう利⽤するのか⾒てい
きます。
図 2.18に、上に書いた Aから Hまでの符号に対するハフマン⽊を⽰してい

ます。葉の重みは、Aが相対頻度 8、Bが相対頻度 3、ほかの⽂字は相対頻度 1
で現れるようなメッセージに対してこの⽊が設計されているということを⽰し
ています。
ハフマン⽊があれば、どの記号に対する符号も、根から始めてその記号を

持つ葉にたどり着くまで下りていくことで知ることができます。左の枝を下り
るたびに符号に 0を追加し、右の枝を下りるたびに 1を追加します (どちらの
枝を下りるかについては、どちらの枝にその記号の葉があるか、またはその記
号を集合に含むかを⾒て決めます)。例えば、図 2.18の根から始めて Dの葉に
着くまでには、右枝、左枝、右枝、右枝とたどっていきます。そのため、Dに
対する符号は 1011となります。
ハフマン⽊を使ってビット列を復号するには、根から始めて、左と右のど

ちらの枝を下りるのかをビット列の 0と 1の列を使って決めます。葉に着くた
びに、メッセージ中の記号を新たにひとつ⽣成し、そこでまた根に戻って次の
記号を探します。例えば、上記の⽊と 10001010というビット列があるとしま
す。根から始めて、右の枝を下り (ビット列の最初のビットは 1なので)、それ
から左の枝を下り (⼆番⽬のビットは 0なので)、次に左の枝を下ります (三番

⽬のビットも 0なので)。こうすると Bの葉にたどり着くので、復号したメッ
セージの最初の記号は Bです。ここでまた根から始めて、ビット列中の次のビ
ットは 0 なので、左に動きます。こうすると、Aの葉に着きます。それからま
た根から始めると、残りのビット列は 1010なので、右、左、右、左と動いて C
に着きます。つまり、メッセージ全体は BACということになります。

ハフマン⽊の⽣成
記号の “アルファベット”と相対頻度がわかっているとき、“最善の”符号を作
るにはどうしたらいいでしょうか (⾔い換えると、メッセージを最⼩のビット
数で符号化するのはどのような⽊でしょうか)。ハフマンはこれを⾏うアルゴ
リズムを与え、結果となる符号が実際に、符号構築時に使った頻度と記号の相
対頻度が⼀致するようなメッセージを符号化するのに最適な可変⻑符号である
ということを⽰しました。ここではハフマン符号の最適性の証明はせずに、ハ
フマン⽊の構築の仕⽅のみを⽰します。42

ハフマン⽊を⽣成するアルゴリズムはとても単純です。考え⽅としては、出
現頻度が最低の記号が根から⼀番遠いところに出てくるように⽊を配置すると
いうものです。まず、符号構築対象の初期データによって決まる、記号と頻度
を持つ葉ノードの集合から始めます。ここで、重みが⼩さいほうから⼆つの葉
を選び、⼆つをくっつけて新しいノードを作り、新しいノードの左と右の枝が
その⼆つのノードになるようにします。新しいノードの重みは⼆つの重みの合
計とします。元の集合からその⼆つの葉を取り除き、代わりに新しいノードを
⼊れます。それからこのプロセスを続けていきます。各段階で、重みが⼩さい
ほうから⼆つのノードを集合から取り除き、それらを左と右の枝として持つノ
ードを代わりに⼊れることよって、その⼆つのノードをくっつけていきます。
ノードがひとつだけになったら、それが⽊全体の根となり、プロセスは停⽌し
ます。以下に、図 2.18のハフマン⽊が⽣成される過程を⽰します。

Initial leaves {(A 8) (B 3) (C 1) (D 1) (E 1) (F 1) (G 1) (H 1)}
Merge {(A 8) (B 3) ({C D} 2) (E 1) (F 1) (G 1) (H 1)}
Merge {(A 8) (B 3) ({C D} 2) ({E F} 2) (G 1) (H 1)}
Merge {(A 8) (B 3) ({C D} 2) ({E F} 2) ({G H} 2)}
Merge {(A 8) (B 3) ({C D} 2) ({E F G H} 4)}
Merge {(A 8) ({B C D} 5) ({E F G H} 4)}
Merge {(A 8) ({B C D E F G H} 9)}

42ハフマン符号の数学的性質についての議論はHamming 1980を参照。

Final merge {({A B C D E F G H} 17)}

このアルゴリズムは常に同じ⽊になるとは限りません。各ステップで、最⼩の
重みとなるノードのペアが⼀意であるとは限らないからです。また、⼆つのノ
ードをくっつける順番をどうするか (つまり、どちらを右枝としてどちらを左
枝とするか)も任意です。

ハフマン⽊の表現
以下の練習では、ハフマン⽊を使ってメッセージの暗号化・復号化を⾏ったり、
上で概要を⽰したアルゴリズムに従ってハフマン⽊を作ったりするシステムを
扱います。まず、⽊をどのように表現するかを検討しましょう。
⽊の葉は、leafという記号、その葉の記号、重みからなるリストによって

表現します。
(define (make-leaf symbol weight) (list 'leaf symbol weight))
(define (leaf? object) (eq? (car object) 'leaf))
(define (symbol-leaf x) (cadr x))
(define (weight-leaf x) (caddr x))

⼀般の⽊は、左枝、右枝、記号の集合、重みからなります。記号の集合は、単
純に記号のリストにして、より複雑な集合の表現は使わないことにします。⼆
つのノードをくっつけて⽊を作るとき、その⽊の重みは⼆つのノードの重みの
合計となり、記号の集合は、⼆つのノードが持つ記号の集合の和集合となりま
す。記号の集合はリストとして表されているため、和集合は2.2.1節で定義した
append ⼿続きによって作ることができます。
(define (make-code-tree left right)

(list left
right
(append (symbols left) (symbols right))
(+ (weight left) (weight right))))

⽊をこうやって作ると、セレクタは次のようになります。
(define (left-branch tree) (car tree))
(define (right-branch tree) (cadr tree))
(define (symbols tree)

(if (leaf? tree)

(list (symbol-leaf tree))
(caddr tree)))

(define (weight tree)
(if (leaf? tree)

(weight-leaf tree)
(cadddr tree)))

⼿続き symbolsと weightは、渡されたものが葉である場合と⼀般の⽊で
ある場合とで、少し違うことをしなければいけません。これらは、ジェネリッ
ク⼿続き (generic procedure)(⼆種類以上のデータを扱う⼿続き)の簡単な例に
なっています。これについては、2.4節と2.5節で詳しく説明します。

復号化⼿続き
以下の⼿続きは、復号化アルゴリズムを実装したものです。0と 1のリストに
加え、ハフマン⽊を引数に取ります。
(define (decode bits tree)

(define (decode-1 bits current-branch)
(if (null? bits)

'()
(let ((next-branch

(choose-branch (car bits) current-branch)))
(if (leaf? next-branch)

(cons (symbol-leaf next-branch)
(decode-1 (cdr bits) tree))

(decode-1 (cdr bits) next-branch)))))
(decode-1 bits tree))

(define (choose-branch bit branch)
(cond ((= bit 0) (left-branch branch))

((= bit 1) (right-branch branch))
(else (error "bad bit: CHOOSE-BRANCH" bit))))

⼿続き decode-1は、残りのビットのリストと、⽊の中での現在位置という、⼆
つの引数を取ります。リストの次のビットが 0か 1かによって、左か右の枝
を “下りて”いきます (これは⼿続き choose-branchで⾏っています)。葉にた
どり着くと、⽊の根から始めてメッセージの残りを復号した結果に葉の記号を
consすることで、メッセージの次の記号として返します。choose-branchの最

終節のエラーチェックに注意してください。ここでは、⼊⼒データの中の 0ま
たは 1以外のものを⼿続きが⾒つけた場合にエラーを出します。

重みつき要素の集合
この⽊の表現では、葉でないノードはそれぞれ記号の集合を持つようになって
いて、その集合は単純なリストとして表現してきました。しかし、上で検討し
た⽊の⽣成アルゴリズムでは、葉と⽊の集合も扱って、⼩さいほうから⼆つの
項⽬を順にくっつけていくという必要がありました。集合の最⼩項⽬を何度も
求めることになるので、この種の集合は順序つきの表現を使うほうが便利です。
葉と⽊の集合は、重みの昇順に配置した要素のリストとして表現すること

にします。以下の adjoin-set⼿続きは集合を構築するためのものですが、こ
れは 練習問題 2.61で説明したものと似ています。しかし、項⽬の⽐較は重みに
よって⾏い、また追加する要素がすでに集合にあるということはありません。
(define (adjoin-set x set)

(cond ((null? set) (list x))
((< (weight x) (weight (car set))) (cons x set))
(else (cons (car set)

(adjoin-set x (cdr set))))))

次の⼿続きは、((A 4) (B 2) (C 1) (D 1))のような記号と頻度のペアのリス
トを取り、初期状態での葉の順序つき集合を構築し、ハフマンアルゴリズムに
従ってすぐ連結できるようにします。
(define (make-leaf-set pairs)

(if (null? pairs)
'()
(let ((pair (car pairs)))

(adjoin-set (make-leaf (car pair) ; symbol
(cadr pair)) ; frequency

(make-leaf-set (cdr pairs))))))

練習問題 2.67: 符号化⽊とサンプルメッセージを次のように定義
する。
(define sample-tree

(make-code-tree (make-leaf 'A 4)
(make-code-tree

(make-leaf 'B 2)
(make-code-tree
(make-leaf 'D 1)
(make-leaf 'C 1)))))

(define sample-message '(0 1 1 0 0 1 0 1 0 1 1 1 0))

decode⼿続きを使ってメッセージを復号し、結果を⽰せ。
練習問題 2.68: encode⼿続きは、引数としてメッセージと⽊を取
り、メッセージを符号化したビットのリストを返す。
(define (encode message tree)

(if (null? message)
'()
(append (encode-symbol (car message) tree)

(encode (cdr message) tree))))

encode-symbolは、与えられた⽊に従って与えられた記号を符号
化したビット列を返す。この⼿続きを書け。記号が⽊に含まれて
いなければエラーを出すように encode-symbolを設計せよ。練習
問題 2.67で得た結果をサンプルの⽊を使って符号化し、それが元
のサンプルメッセージと同じになるかどうか確認して、書いた⼿
続きをテストせよ。
練習問題 2.69: 以下の⼿続きは、引数として記号・頻度ペアのリ
スト (同じ記号が⼆つ以上のペアに出てくることはない) を取り、
ハフマンアルゴリズムに従ってハフマン符号化⽊を⽣成する。
(define (generate-huffman-tree pairs)

(successive-merge (make-leaf-set pairs)))

make-leaf-setは、上で記述した、ペアのリストを葉の順序つき
集合に変換する⼿続きである。successive-mergeは、集合の中で
重みが最⼩の要素を make-code-treeを使って順番にくっつけて
いき、最後に要素がひとつだけ残るようにするというものである。
その要素が求めるハフマン⽊となる。この⼿続きを書け (この⼿続
きにはちょっと厄介なところがあるが、そこまで複雑ではない。も
し⼿続きの設計が複雑になったとしたら、ほぼ確実に何かを間違
えている。順序つきの集合表現を使っているということが⼤きな
助けになる)。

練習問題 2.70: 以下に⽰す 8記号の相対頻度つきアルファベット
は、1950年代のロックの歌詞を効率よく符号化できるよう設計さ
れたものである (“アルファベット”の “記号”が単独の⽂字である
とは限らないことに注意)。
A 2 GET 2 SHA 3 WAH 1
BOOM 1 JOB 2 NA 16 YIP 9

generate-huffman-tree(練習問題 2.69)を使って対応するハフマ
ン⽊を⽣成し、encode (練習問題 2.68)を使って次のメッセージを
符号化せよ。
Get a job
Sha na na na na na na na na
Get a job
Sha na na na na na na na na
Wah yip yip yip yip yip yip yip yip yip
Sha boom

符号化には何ビット必要だろうか。もし固定⻑符号をこの 8記号
アルファベットに使ったとしたら、最低でどれだけのビット数が
必要になるだろうか。
練習問題 2.71: n記号のアルファベットに対するハフマン⽊があ
り、記号の相対頻度は 1, 2, 4, . . . , 2n−1であるとする。n = 5、n = 10

の場合の⽊をスケッチせよ。そのような⽊では、(⼀般の nについ
て)最も頻度の⾼い記号を符号化するのに何ビット必要になるだ
ろうか。最も頻度の低い記号はどうだろうか。
練習問題 2.72: 練習問題 2.68で設計した符号化⼿続きについて考
える。ひとつの記号を符号化するのに必要なステップ数の増加オ
ーダーはどのようになるだろうか。各ノードに着くたびに記号リ
ストを検索するのに必要なステップ数を含めることを忘れないよ
うに。この問題の⼀般の場合について答えることは難しい。ここ
では、n記号の相対頻度が 練習問題 2.71のようになっている特別
な場合について考えよう。アルファベット中で頻度が最⼤の記号
と最⼩の記号を符号化するのにかかるステップ数の増加オーダー
を (nの関数として)答えよ。

2.4 抽象データの多重表現
ここまで、データ抽象化というものを紹介してきました。これはシステムを構
築するための⽅法論で、プログラムの操作対象となるデータオブジェクトに対
してどのような実装を選ぶかということと独⽴に、プログラムのほとんどの部
分を記述できるようにするというものでした。例えば2.1.1節では、有理数を使
うプログラムを設計するというタスクを、複合データを構築するためにコンピ
ュータ⾔語が持っている基本メカニズムによって有理数を実装するというタス
クから切り離すやり⽅について⾒てきました。ポイントとなる考え⽅は、抽象
化の壁—この場合は、有理数のセレクタとコンストラクタ (make-rat, numer,
denom)—を建てて、有理数をどう使うかというところと、有理数のリスト構造
による表現という内部の構造とを分離するということでした。結果として、プ
ログラムは図 2.1に⽰したような構造になりました。
これらのデータ抽象化の壁は、複雑性をコントロールする強⼒な道具です。

データオブジェクトの内部表現を分離することで、⼤きなプログラムの設計と
いうタスクを、別々に⾏うことのできる⼩さなタスクの集まりに分割できま
す。しかし、この⼿のデータ抽象化はまだ⼗分に強⼒とは⾔えません。という
のは、データ構造の “内部表現”というものがいつも意味があるとは限らない
からです。
⼀つ⽬には、データオブジェクトの便利な表現⽅法が⼆つ以上あって、複

数の表現を扱えるようなシステムを設計したいという場合があります。単純な
例について考えると、複素数は⼆種類のほぼ等価な⽅法、直交形式 (実部と虚
部)と極形式 (絶対値と偏⾓)によって表すことができます。直交形式のほうが
適切な場合もあれば、極形式のほうが適切な場合もあります。実際に、複素数
が両⽅の形式で表され、複素数を操作する⼿続きはどちらか⼀⽅の表現を扱う
といったシステムも、⼗分に考えられます。
さらに重要なことは、プログラミングシステムは仕様変更を繰り返しなが

ら⻑期間にわたって多くの⼈々によって設計されるということがよくあるとい
うことです。そのような環境では、データ表現として何を選ぶかについてみん
なが前もって合意するということはまずありません。つまり、データの表現を
それを使うところから分離するデータ抽象化の壁に加えて、複数のデータ表現
をそれぞれ分離し、ひとつのプラットフォーム上でそれらを共存させる抽象化
の壁が必要だということです。また、⼤きなプログラムというものは、別々に
設計されたすでに存在するモジュールを組み合わせて作られることが多いの
で、プログラマがモジュールを加法的 (additively)に、つまり、これらのモジュ
ールを設計し直したり実装し直したりすることなく、より⼤きなシステムに組

add-complex  sub-complex  mul-complex  div-complex

複素数を使うプログラム

複素数算術演算パッケージ

直交形式
による表現

極形式
による表現

リスト構造とマシンの基本算術演算

図 2.19: 複素数システムのデータ抽象化の壁

み込んでいくための規約が必要になります。
この節では、プログラムの部品によって違う⽅法で表現されるデータを

扱うやり⽅について学びます。そのためには、ジェネリック⼿続き (generic
procedure)— ⼆種類以上の⽅法で表現されるデータを扱える⼿続き—を構築す
る必要があります。ジェネリック⼿続きを構築するためにここで使う主なテク
ニックは、タイプタグ (type tag)を持つデータオブジェクト、つまりどのよう
に処理されるべきかという情報を明⽰的に持っているデータオブジェクトを使
うということです。また、データ主導 (data-directed)プログラミングについて
も検討します。これは、ジェネリック演算によって加法的にシステムを組み⽴
てていくための強⼒で便利な実装戦略です。
最初に、単純な複素数の例を扱います。タイプタグとデータ主導スタイル

を使うことで、“複素数”データオブジェクトという抽象概念を維持したまま、
複素数の表現として直交形式と極形式を別々に設計できるということを⾒てい
きます。これは、複素数がどのように表現されているかということと独⽴に複
素数の各部分にアクセスするジェネリックセレクタによって複素数の算術演算
(add-complex, sub-complex, mul-complex, div-complex)を定義することで達
成できます。これによってできる複素数システムでは、図 2.19に⽰すように、
⼆種類の異なる抽象化の壁を持つことになります。“⽔平な”抽象化の壁は、図
2.1に出てきたものと同じ役割を果たします。それらは、“⾼レベル”演算を “低
レベル”演算から分離するものです。それに加えて、複数の表現を別々に設計
して組み込めるようにするための “垂直な”壁があります。

2.5節では、タイプタグとデータ主導スタイルを使ってジェネリック算術

演算パッケージを開発するやり⽅を⽰します。このシステムは、すべての種類
の “数値”を操作するのに使える⼿続き (add, mul など)を提供し、新たな種類
の数値が必要になったら簡単に拡張できるようになっています。2.5.3節では、
記号代数を実⾏するシステムの中でジェネリック数値演算を使う⽅法を⽰し
ます。

2.4.1 複素数の表現
ここでは、ジェネリック演算を使うプログラムの単純で⾮現実的な例として、
複素数に対する算術演算を⾏うシステムを開発していきます。まず、複素数を
順序つきペアとして表すときに考えられる⼆つの表現⽅法、直交形式 (実部と
虚部)と極形式 (絶対値と偏⾓)について検討します。43 2.4.2節では、タイプタ
グとジェネリック演算を使うことによって両⽅の表現を共存させるやり⽅を⽰
します。
有理数と同じように、複素数は順序つきペアとして⾃然に表現できます。複

素数の集合は、“実”軸と “虚”軸という⼆つの直交する軸を持つ⼆次元空間と
考えることができます (図 2.20参照)。この⾒⽅をすると、z = x+ iy (i2 = −1)
という複素数は、実座標が xで虚座標が yの平⾯上の点として考えることがで
きます。複素数の和は、この表現では、座標の和になります。

実部(z1 + z2) = 実部(z1) + 実部(z2),

虚部(z1 + z2) = 虚部(z1) + 虚部(z2).

複素数をかけ算するときには、複素数を極形式で、つまり⼤きさと⾓度 (
図 2.20内の rと A)を使って考えるほうが⾃然です。⼆つの複素数の積は、ひ
とつの複素数をもう⼀⽅の⻑さをかけて伸縮させ、もう⼀⽅の⾓度だけ回転さ
せて得られるベクトルになります。

⼤きさ(z1 · z2) = ⼤きさ(z1) · ⼤きさ(z2),

⾓度(z1 · z2) = ⾓度(z1) + ⾓度(z2).

つまり、複素数には⼆つの異なる表現があり、それぞれ異なる演算に適してい
るということです。しかし、複素数を使うプログラムを書いている⼈の視点か

43実際の計算システムでは、ほとんどの場合、直交形式が選ばれます。これは、直交形
式と極形式の変換時に発⽣する丸め誤差のためです。このため、この複素数の例は現実
的ではありません。しかし、ジェネリック演算を使ったシステムの設計を説明する例と
しては明快ですし、この章の後のほうで開発する、より実質的なシステムへの導⼊とし
ても役に⽴ちます。

184



虚部

実部

z = x + iy = reiA

A

y

x

r

図 2.20: 平⾯上の点としての複素数

らは、データ抽象化の原則により、コンピュータがどちらの表現を使っていて
も、複素数の演算すべてが使えるようになっていなければいけません。例えば、
直交形式によって指定された複素数の絶対値が求められると便利なことがよく
あります。同じように、極形式によって指定された複素数の実部を求められる
と便利なことがよくあります。
そのようなシステムを設計するにあたっては、2.1.1節で有理数パッケージ

を設計する際に使ったのと同じデータ抽象化戦略が使えます。複素数演算は、
real-part, imag-part, magnitude, angleという四つのセレクタによって実装
されているとします。また、複素数の構築には⼆つの⼿続きがあるとします。
make-from-real-imagは、指定された実部と虚部を持つ複素数を返し、make-
from-mag-angは、指定された絶対値と偏⾓を持つ複素数を返します。これら
の⼿続きは、任意の複素数に対して
(make-from-real-imag (real-part z) (imag-part z))

と
(make-from-mag-ang (magnitude z) (angle z))

の両⽅とも zに等しい複素数を返すという性質を持っています。
これらのコンストラクタとセレクタを使うことで、2.1.1節で有理数につい

て⾏ったように、コンストラクタとセレクタによって規定される “抽象データ”
を使って複素数の算術演算を実装できます。上の式に⽰したように、⾜し算と

引き算は実部と虚部を使って、またかけ算と割り算は絶対値と偏⾓を使って⾏
えます。
(define (add-complex z1 z2)

(make-from-real-imag (+ (real-part z1) (real-part z2))
(+ (imag-part z1) (imag-part z2))))

(define (sub-complex z1 z2)
(make-from-real-imag (- (real-part z1) (real-part z2))

(- (imag-part z1) (imag-part z2))))
(define (mul-complex z1 z2)

(make-from-mag-ang (* (magnitude z1) (magnitude z2))
(+ (angle z1) (angle z2))))

(define (div-complex z1 z2)
(make-from-mag-ang (/ (magnitude z1) (magnitude z2))

(- (angle z1) (angle z2))))

複素数パッケージを完成させるには、複素数の表現を選び、コンストラクタと
セレクタを基本数値と基本リストを使って実装しなければいけません。すぐわ
かるように、この⽅法は⼆通りあります。“直交形式”のペア (実部、虚部)とし
て表現するか、“極形式”のペア (絶対値、偏⾓)として表現するかです。どち
らを選ぶべきでしょうか。
この⼆つの選択の具体的な例として、Ben Bitdiddleと Alyssa P. Hackerと

いう、独⽴に複素数システムの表現を設計している⼆⼈のプログラマがいると
想像してください。Ben は複素数を直交形式で表現することにします。この
場合、複素数の実部と虚部を選択することや、複素数を与えられた実部と虚部
から構築することは、素直に実現できます。絶対値と偏⾓を求めたり、複素数
を与えられた絶対値と偏⾓から構築するのは、次のような三⾓法の関係を使い
ます。

x = r cosA, r =
√
x2 + y2,

y = r sinA, A = arctan(y, x),

これは、実部と虚部 (x, y)と絶対値と偏⾓ (r,A)の関係を表しています。44 こ
のため、Benの表現は以下のようなセレクタとコンストラクタによって書くこ
とができます。

44ここで出てきた逆正接関数は Schemeの atan⼿続きで計算できるもので、引数 y, x
を取り、正接が y/xとなる⾓度を返すよう定義されています。引数の符号によって⾓度
の象限が決まります。

(define (real-part z) (car z))
(define (imag-part z) (cdr z))
(define (magnitude z)

(sqrt (+ (square (real-part z))
(square (imag-part z)))))

(define (angle z)
(atan (imag-part z) (real-part z)))

(define (make-from-real-imag x y) (cons x y))
(define (make-from-mag-ang r a)

(cons (* r (cos a)) (* r (sin a))))

⼀⽅、Alyssaは複素数を極形式で表現することにしました。彼⼥の⽅法では、
絶対値と偏⾓のセレクトは素直にできます。しかし、実部と虚部を求めるには、
三⾓法の関係を使う必要があります。Alyssaの表現は次のようになります。
(define (real-part z) (* (magnitude z) (cos (angle z))))
(define (imag-part z) (* (magnitude z) (sin (angle z))))
(define (magnitude z) (car z))
(define (angle z) (cdr z))
(define (make-from-real-imag x y)

(cons (sqrt (+ (square x) (square y)))
(atan y x)))

(define (make-from-mag-ang r a) (cons r a))

データ抽象化の規律に従っているため、add-complex, sub-complex, mul-
complex, div-complex の同じ実装が、Ben の表現と Alyssa の表現のどちら
に対してもうまく動くということが保証されます。

2.4.2 タグつきデータ
データ抽象化は、“最⼩責任の原則”の応⽤として⾒ることもできます。2.4.1
節で複素数システムを実装したとき、Benの直交形式による表現と Alyssaの
極形式による表現のどちらを使うこともできました。セレクタとコンストラク
タが抽象化の壁を作っているため、データオブジェクトの具体的な表現に何を
選ぶかということを最後の瞬間まで遅らせることができ、そのためシステム設
計の柔軟性を最⼤限にできます。
最⼩責任の原則は、さらに極限まで進めることができます。もし望むなら、

セレクタとコンストラクタを設計した “後”になっても表現を曖昧にしたまま

で、Benの表現と Alyssaの表現の “両⽅”を使うようにすることもできます。
しかし、両⽅の実装が単独のシステムに含まれている場合、極形式のデータと
直交形式のデータを区別するための何らかの⽅法が必要になります。そうしな
いと、例えば (3, 4)というペアの magnitude (絶対値)を求めるよう⾔われた場
合に、答えを (数値を直交形式と解釈して)5とするべきか (数値を極形式と解
釈して)3とするべきかわかりません。これを区別できるようにする素直なやり
⽅は、タイプタグ (type tag)—rectangularまたは polarという記号—を、そ
れぞれの複素数の⼀部として含めるというものです。こうすると、複素数を操
作することが必要になったときに、どちらのセレクタを適⽤するべきかをこの
タグを使って決めることができます。
タグつきデータを操作するために、type-tagと contentsという⼿続きを

持っていると想定します。type-tagはデータオブジェクトからタグを抽出す
るもので、contentsは実際の中⾝ (複素数の場合、極形式または直交形式の座
標)を抽出するものです。また、attach-tagという⼿続きも持っていると仮定
します。これは、タグと中⾝を取り、タグつきデータオブジェクトを返すとい
うものです。これを実装する素直なやり⽅は、普通のリスト構造を使うという
ものです。
(define (attach-tag type-tag contents)

(cons type-tag contents))
(define (type-tag datum)

(if (pair? datum)
(car datum)
(error "Bad tagged datum: TYPE-TAG" datum)))

(define (contents datum)
(if (pair? datum)

(cdr datum)
(error "Bad tagged datum: CONTENTS" datum)))

これらの⼿続きを使って、直交形式と極形式をそれぞれ識別する述語 rectangular?,
polar?を定義できます。
(define (rectangular? z)

(eq? (type-tag z) 'rectangular))
(define (polar? z) (eq? (type-tag z) 'polar))

タイプタグを使うことにしたので、Benと Alyssaはコードを修正して、異なる
表現が同じシステムの中で共存できるようにしました。Benが複素数を構築す

るたびに、それに直交形式というタグをつけます。Alyssaが複素数を構築する
たびに、それに極形式というタグをつけます。それに加えて、Benと Alyssaは
それぞれの⼿続きの名前が衝突しないようにしなければいけません。これを⾏
う⽅法としては、Benが⾃分の表現の⼿続きに接尾辞 rectangularを追加し、
Alyssaは⾃分の表現⼿続きに polarを追加するというものがあります。以下
は、2.4.1節の Benの直交形式表現を修正したものです。
(define (real-part-rectangular z) (car z))
(define (imag-part-rectangular z) (cdr z))
(define (magnitude-rectangular z)

(sqrt (+ (square (real-part-rectangular z))
(square (imag-part-rectangular z)))))

(define (angle-rectangular z)
(atan (imag-part-rectangular z)

(real-part-rectangular z)))
(define (make-from-real-imag-rectangular x y)

(attach-tag 'rectangular (cons x y)))
(define (make-from-mag-ang-rectangular r a)

(attach-tag 'rectangular
(cons (* r (cos a)) (* r (sin a)))))

Alyssaの極形式表現の修正版のは次のようになりました。
(define (real-part-polar z)

(* (magnitude-polar z) (cos (angle-polar z))))
(define (imag-part-polar z)

(* (magnitude-polar z) (sin (angle-polar z))))
(define (magnitude-polar z) (car z))
(define (angle-polar z) (cdr z))
(define (make-from-real-imag-polar x y)

(attach-tag 'polar
(cons (sqrt (+ (square x) (square y)))

(atan y x))))
(define (make-from-mag-ang-polar r a)

(attach-tag 'polar (cons r a)))

どちらのジェネリックセレクタも、引数のタグをチェックして、そのタイプのデ
ータを扱う適切な⼿続きを呼ぶ⼿続きとして実装されています。例えば、複素数

189



の実部を得る場合、real-partはタグを調べ、Benの real-part-rectangular
と Alyssaの real-part-polarのどちらを使うのかを決定します。どちらの場
合でも、contentsを使って⽣のタグなしデータを取り出し、それを必要に応じ
て直交形式または極形式の⼿続きに渡します。
(define (real-part z)

(cond ((rectangular? z)
(real-part-rectangular (contents z)))

((polar? z)
(real-part-polar (contents z)))

(else (error "Unknown type: REAL-PART" z))))
(define (imag-part z)

(cond ((rectangular? z)
(imag-part-rectangular (contents z)))

((polar? z)
(imag-part-polar (contents z)))

(else (error "Unknown type: IMAG-PART" z))))
(define (magnitude z)

(cond ((rectangular? z)
(magnitude-rectangular (contents z)))

((polar? z)
(magnitude-polar (contents z)))

(else (error "Unknown type: MAGNITUDE" z))))
(define (angle z)

(cond ((rectangular? z)
(angle-rectangular (contents z)))

((polar? z)
(angle-polar (contents z)))

(else (error "Unknown type: ANGLE" z))))

複素数の算術演算を実装するのには、2.4.1 節の⼿続き add-complex, sub-
complex, mul-complex, div-complexと同じものが使えます。それらが呼び出
すセレクタはジェネリックなので、どちらの表現を使っても動くからです。例
えば、⼿続き add-complexは相変わらずそのままで、以下のようになります。
(define (add-complex z1 z2)

(make-from-real-imag (+ (real-part z1) (real-part z2))
(+ (imag-part z1) (imag-part z2))))

図 2.21: ジェネリックな複素数算術演算システムの構造

最後に、複素数を構築するのに Benの表現と Alyssaの表現のどちらを使うか
を決める必要があります。実部と虚部が渡されたときには直交形式を使い、⼤
きさと⾓度を渡されたときには極形式を使うというのが妥当でしょう。
(define (make-from-real-imag x y)

(make-from-real-imag-rectangular x y))
(define (make-from-mag-ang r a)

(make-from-mag-ang-polar r a))

このようにして作った複素数システムは、図 2.21に⽰すような構造を持ちます。
システムは、⽐較的独⽴した三つの部分に分解されています。複素数算術演算、
Alyssaの極形式による実装、Benの直交形式による実装です。極形式と直交形
式の実装は、Benと Alyssaが別々に書いていたとしても問題ありません。第三
者のプログラマが抽象コンストラクタ、セレクタによって複素数算術演算⼿続
きを実装する際に、内部表現として両⽅とも使うことができます。
それぞれのデータオブジェクトには型のタグがついているので、セレクタ

はジェネリックにデータを操作します。つまり、それぞれのセレクタは適⽤さ
れるデータの個々の型によって異なるふるまいをするように定義されている
ということです。別々の表現をインターフェイスで接続する際の⼀般的なメカ
ニズムに注⽬してください。ある表現の実装 (例えば、Alyssaの極形式パッケ
ージ)の中で極形式型の複素数を操作するときには、タグを取り除き、中⾝を

Alyssaのコードに渡します。逆に、Alyssaが⼀般使⽤のための数値を構築する
際には、それに型のタグをつけて、上のレベルの⼿続きが適切に認識できるよ
うにします。データオブジェクトをレベル間でやりとりする際にタグをつけた
り外したりするというこの規律は、重要な組織化戦略になりえます。2.5節で
は、このことについて⾒ていきます。

2.4.3 データ主導プログラミングと加法性
データの型をチェックし適切な⼿続きを呼ぶ⼀般的な戦略は型によるディスパ
ッチ (dispatching on type)と呼ばれるものです。これは、システム設計でモジ
ュール性を達成するための強⼒な戦略です。⼀⽅で、2.4.2節のようにディスパ
ッチを実装することには⼆つの⼤きな弱点があります。弱点のひとつは、ジェ
ネリックインターフェイス⼿続き (real-part, imag-part, magnitude, angle)
がすべての異なる表現について知っていなければいけないということです。例
えば、複素数の新しい表現をこの複素数システムに組み込みたいとします。そ
のためには、この新しい表現を型で識別できるようにして、それから各ジェネ
リックインターフェイス⼿続きにその新しい型をチェックする節を追加し、そ
してその表現に対する適切なセレクタを適⽤することが必要になるでしょう。
このテクニックのもうひとつの弱点は、個々の表現は別々に設計できるも

のの、システム全体で⼆つの⼿続きが同じ名前を持つということがないように
保証しなければならないということです。このために、Benと Alyssaは2.4.1
節の元の⼿続きの名前を変える必要があったのでした。
この⼆つの弱点の元となっているのは、ジェネリックインターフェイスを

実装するテクニックが 加法的 (additive)でないということです。ジェネリック
セレクタ⼿続きを実装する⼈は、新しい表現が組み込まれるたびにそれらの⼿
続きを変更しなければならず、また個々の表現のインターフェイスを作る⼈は、
名前の衝突を避けるためにコードを変更しなければなりません。どちらの場合
でも、コードに加えるべき変更は単純なものですが、それでもやはり⾏わなけ
ればならず、このことが不便さとエラーの原因になります。これは複素数シス
テムに関しては現時点ではあまり⼤きな問題ではありませんが、仮に複素数の
表現が⼆つではなく数百個あったとしたらどうでしょうか。また、抽象データ
インターフェイスにはメンテナンスするべきジェネリックセレクタが数多くあ
るとしたらどうでしょうか。また、すべてのインターフェイス⼿続きやすべて
の表現について知っているプログラマが誰もいないとしたらどうでしょうか。
この問題は現実的なもので、⼤規模データベース管理システムのようなプログ
ラムではこの問題に取り組む必要があります。

図 2.22: 複素数システム演算のテーブル

ここで、システム設計をさらにモジュール化するための⼿段が必要になり
ます。データ主導プログラミング (data-directed programming)として知られる
プログラミングテクニックはその⼿段を提供してくれます。データ主導プログ
ラミングがどうやって動くのかを理解するために、まず次のことに注⽬します。
いろいろな型の集合に対して共通のものとなるジェネリック演算を扱う際には
いつも、実質的には、ひとつの軸が可能な演算で、もうひとつの軸が可能な型
であるような⼆次元のテーブルを扱っていることになります。テーブルの項⽬
は、それぞれの引数の型に対するそれぞれの演算を実装する⼿続きです。前の
節で開発した複素数システムでは、演算の名前、データタイプ、実際の⼿続き
の間の対応関係は、ジェネリックインターフェイス⼿続きのいろいろな条件節
の中に広がっていました。しかし、それと同じ情報を図 2.22に⽰すようにテー
ブルの中に構造化することもできたところです。
データ主導プログラミングは、そのようなテーブルを直接扱うようにプロ

グラムを設計するテクニックです。前のほうで、複素数算術演算のコードと、
それぞれ型によって明⽰的なディスパッチを⾏う⼿続きの集合となっている⼆
つの表現パッケージとの間の橋渡しをするメカニズムを実装しました。ここで
は、そのインターフェイス部分を、演算の名前と引数の型の組み合わせをテー
ブルから検索して適⽤するべき正しい⼿続きを⾒つけ、それを引数の中⾝に適
⽤するということを⾏う、単⼀の⼿続きとして実装します。こうすると、新し
い表現パッケージをシステムに追加する際に、既存の⼿続きには何の変更も加
えなくても⼤丈夫です。テーブルに新しい項⽬を追加するだけで⼗分です。
この計画を実装するために、演算-型テーブルを操作する putと getという

⼆つの⼿続きがあるということにします。
• (put ⟨op ⟩ ⟨type⟩ ⟨item⟩)は、テーブルの ⟨op ⟩と ⟨type⟩が指すところに

⟨item⟩を⼊れる。
• (get ⟨op ⟩ ⟨type⟩)は、テーブルから ⟨op⟩, ⟨type⟩の項⽬を検索し、そこ
で⾒つかった項⽬を返す。⾒つからなければ、getは falseを返す。

今のところは、putと getは⾔語に含まれているということにします。第 3章
(3.3.3節)では、これらを含め、テーブル操作演算の実装⽅法について学びます。
それでは、複素数システムでデータ主導プログラミングをどう使うかにつ

いて⾒ていきましょう。直交形式表現を開発した Benは、元のものと同じよう
にコードを実装します。それから⼿続きの集合、つまりパッケージ (package)
を定義し、システムに直交形式の数値の扱い⽅を知らせるテーブルに項⽬を追
加することで、これらの⼿続きをシステムのほかの部分と接続します。これは、
次の⼿続きを呼ぶことで⾏えます。
(define (install-rectangular-package)

;; 内部⼿続き
(define (real-part z) (car z))
(define (imag-part z) (cdr z))
(define (make-from-real-imag x y) (cons x y))
(define (magnitude z)

(sqrt (+ (square (real-part z))
(square (imag-part z)))))

(define (angle z)
(atan (imag-part z) (real-part z)))

(define (make-from-mag-ang r a)
(cons (* r (cos a)) (* r (sin a))))

;; システムのほかの部分とのインターフェイス
(define (tag x) (attach-tag 'rectangular x))
(put 'real-part '(rectangular) real-part)
(put 'imag-part '(rectangular) imag-part)
(put 'magnitude '(rectangular) magnitude)
(put 'angle '(rectangular) angle)
(put 'make-from-real-imag 'rectangular

(lambda (x y) (tag (make-from-real-imag x y))))
(put 'make-from-mag-ang 'rectangular

(lambda (r a) (tag (make-from-mag-ang r a))))
'done)

ここでの内部⼿続きが、2.4.1節で Benが単独で書いたものと同じ⼿続きである
ことに注⽬してください。これらをシステムのほかの部分と接続するのに、変
更を加える必要はありません。それに、これらの⼿続きの定義は組み込み⼿続
きの内部にあるので、直交形式パッケージの外のほかの⼿続きと名前が衝突す
る⼼配もありません。これらの⼿続きをシステムのほかの部分と接続するため
に、Benは⾃分の real-part ⼿続きを、演算名 real-partと型 (rectangular)
のところに⼊れます。ほかのセレクタも同じようにします。45 このインターフ
ェイスは、システムのほかの部分から使うためのコンストラクタも定義してい
ます。46コンストラクタは Benが内部で定義していたものと同じですが、タグ
をつけるというところが違います。

Alyassaの極形式パッケージも似たようなものになります。
(define (install-polar-package)

;; 内部⼿続き
(define (magnitude z) (car z))
(define (angle z) (cdr z))
(define (make-from-mag-ang r a) (cons r a))
(define (real-part z) (* (magnitude z) (cos (angle z))))
(define (imag-part z) (* (magnitude z) (sin (angle z))))
(define (make-from-real-imag x y)

(cons (sqrt (+ (square x) (square y)))
(atan y x)))

;; システムのほかの部分とのインターフェイス
(define (tag x) (attach-tag 'polar x))
(put 'real-part '(polar) real-part)
(put 'imag-part '(polar) imag-part)
(put 'magnitude '(polar) magnitude)
(put 'angle '(polar) angle)
(put 'make-from-real-imag 'polar

(lambda (x y) (tag (make-from-real-imag x y))))
(put 'make-from-mag-ang 'polar

(lambda (r a) (tag (make-from-mag-ang r a))))

45ここでは、rectangular という記号ではなく (rectangular) というリストを使い、
必ずしも同じ型であるとは限らない複数の引数を持つ⼿続きを持てるようにしています。

46コンストラクタを⼊れる場所の型はリストにしなくても⼤丈夫です。コンストラク
タは必ずある特定の型のオブジェクトを作るのに使われるからです。

'done)

Benと Alyssaは、相⼿のものと同じ名前 (例えば、real-partなど)によって定
義された元々の⼿続きをまだ使っていますが、これらの定義は今では別々の⼿
続きの内部定義 (1.1.8節参照)になっているので、名前の衝突は起こりません。
複素数算術演算セレクタは、apply-genericというジェネリック “演算”⼿

続きを使ってテーブルにアクセスします。これは、ジェネリック演算を引数に
適⽤するものです。apply-genericは、テーブルから演算の名前と引数の型に
対応する箇所を検索し、⼿続きがあればそれを適⽤します。47

(define (apply-generic op . args)
(let ((type-tags (map type-tag args)))

(let ((proc (get op type-tags)))
(if proc

(apply proc (map contents args))
(error

"No method for these types: APPLY-GENERIC"
(list op type-tags))))))

apply-genericを使うと、複素数演算のジェネリックセレクタは以下のように
定義できます。
(define (real-part z) (apply-generic 'real-part z))
(define (imag-part z) (apply-generic 'imag-part z))
(define (magnitude z) (apply-generic 'magnitude z))
(define (angle z) (apply-generic 'angle z))

新しい表現がシステムに追加されてもこれらはまったく変わらないということ
に注意してください。

47apply-genericは、練習問題 2.20で説明したドット末尾記法を使います。それぞれ
のジェネリック演算が取る引数の数は違うかもしれないからです。apply-genericでは、
opは apply-genericの第⼀引数を値として持ち、argsは残りの引数のリストを値とし
て持ちます。
また、apply-genericは applyという基本⼿続きを使っています。この⼿続きは⼿続

きとリストという⼆つの引数を取り、リストの要素を引数として⼿続きを適⽤します。例
えば、
(apply + (list 1 2 3 4))

は 10を返します。

また、このパッケージの外部のプログラムが、実部と虚部や絶対値と偏⾓
から複素数を作る際に使うコンストラクタを抽出することもできます。2.4.2
節と同じように、実部と虚部がある場合は直交形式の複素数を構築し、絶対値
と偏⾓がある場合は極形式の複素数を構築します。
(define (make-from-real-imag x y)

((get 'make-from-real-imag 'rectangular) x y))
(define (make-from-mag-ang r a)

((get 'make-from-mag-ang 'polar) r a))

練習問題 2.73: 2.3.2節では、記号微分を⾏うプログラムについて
説明した。
(define (deriv exp var)

(cond ((number? exp) 0)
((variable? exp)
(if (same-variable? exp var) 1 0))

((sum? exp)
(make-sum (deriv (addend exp) var)

(deriv (augend exp) var)))
((product? exp)
(make-sum (make-product

(multiplier exp)
(deriv (multiplicand exp) var))

(make-product
(deriv (multiplier exp) var)
(multiplicand exp))))

⟨more rules can be added here⟩
(else (error "unknown expression type:

DERIV" exp))))

このプログラムは、微分する式の型によってディスパッチを実⾏
していると捉えることもできる。この場合、データの “タイプタ
グ”は代数演算記号 (+など)で、⾏う演算は deriv ということに
なる。基本的な微分を⾏う⼿続きを次のように書き直すと、プロ
グラムをデータ主導スタイルに変形できる。
(define (deriv exp var)

(cond ((number? exp) 0)
((variable? exp) (if (same-variable? exp var) 1 0))
(else ((get 'deriv (operator exp))

(operands exp) var))))
(define (operator exp) (car exp))
(define (operands exp) (cdr exp))

a. 上で何をしているか説明せよ。⼿続き number?と variable?
は、なぜデータ主導ディスパッチとして取り込むことができ
ないのだろうか。

b. 和と積に対する微分⼿続きと、上記のプログラムで使ってい
るテーブルにそれらを組み込む補助コードを書け。

c. 任意の微分規則 (例えば、練習問題 2.56の指数の微分など)を
選び、それをこのデータ主導システムに組み込め。

d. この単純な代数操作では、式の型は式をまとめる代数演算
⼦となっている。しかし、仮に⼿続きのインデックスを逆に
して、derivでディスパッチを⾏う箇所を次のようにすると
する。
((get (operator exp) 'deriv) (operands exp) var)

これに対応して、微分システムにはどのような変更を加える
必要があるだろうか。

練習問題 2.74: Insatiable Enterprises, Inc.(貪欲エンタープライズ
社)は、世界中に散らばった多数の独⽴事業所からなる分散型複合
企業である。この会社のコンピュータ設備は、どのユーザーから
⾒てもネットワーク全体がひとつのコンピュータのように⾒える
ような巧妙なネットワークインターフェイス⽅式によって相互接
続されたところだ。貪欲社の社⻑は、初めてネットワークの機能
を使って事業所ファイルから管理情報を取り出そうとして愕然と
した。事務所ファイルはすべて Schemeのデータ構造として実装
されているのに、使われている個々のデータ構造は事業所ごとに
違うのだ。事業所⻑会議が急遽開かれ、事業所の独⽴性をこれま
で通りに保ったままで、本部の要求を満たせるようにファイルを
統合する戦略を探ることになった。

そのような戦略をデータ主導プログラミングによって実装する⽅
法を⽰せ。例として、各事業所の⼈事記録は単独のファイルからな
り、従業員の名前をキーとしたレコードの集合を持っているとす
る。集合の構造は事業所ごとに異なる。さらに、各従業員のレコー
ドはそれ⾃⾝が (事業所ごとに異なる構造を持つ)集合で、address
と salaryのような識別⼦をキーとした情報を含んでいるとする。
具体的には、

a. 指定した⼈事ファイルから指定した従業員のレコードを取得
する get-record⼿続きを本部向けに実装せよ。この⼿続き
は、任意の事業所のファイルに適⽤できる必要がある。個々
の事業所のファイルはどのように構造化しなければならない
か説明せよ。具体的には、どのような型情報を提供する必要
があるだろうか。

b. 任意の事業所の⼈事ファイル内の与えられた職員のレコード
から給与情報を返す get-salary ⼿続きを本部向けに実装せ
よ。この演算が動くようにするには、レコードはどのように
構造化しなければならないだろうか。

c. 本部向けに、find-employee-record⼿続きを実装せよ。こ
の⼿続きは、全事業所のファイルから与えられた従業員を検
索し、該当レコードを返す。引数として、従業員名と全事業
所のファイルのリストを取るとせよ。

d. 貪欲社が新しい会社を吸収した場合、新しい⼈事情報を中央
システムに組み⼊れるためにはどのような変更が必要になる
だろうか。

メッセージパッシング
データ主導プログラミングでポイントとなる考え⽅は、プログラム中のジェネ
リック演算を 図 2.22のような演算-型テーブルによって明⽰的に扱うというと
ころです。2.4.2節で使ったプログラミングスタイルでは、それぞれの演算に⾃
分のディスパッチの⾯倒を⾒させることによって、必要となる型ディスパッチ
を構築していました。実質的には、それぞれのジェネリック演算⼿続きがテー
ブルの⾏を表すというやり⽅で、演算-型テーブルを⾏に分解していたことに
なります。

もうひとつの実装戦略は、テーブルを列に分解し、データ型によってディ
スパッチを⾏う “賢い演算”を使うのではなく、演算の名前によってディスパ
ッチを⾏う “賢いデータオブジェクト”を使うというものです。これを実現す
るには、データオブジェクト (例えば直交形式の複素数)を⼿続きとして表現
し、その⼿続きは実⾏するべき演算名を引数として取り、指定された演算を実
⾏するというようにします。このような規律に従うと、make-from-real-imag
は以下のように書くことができます。
(define (make-from-real-imag x y)

(define (dispatch op)
(cond ((eq? op 'real-part) x)

((eq? op 'imag-part) y)
((eq? op 'magnitude) (sqrt (+ (square x) (square y))))
((eq? op 'angle) (atan y x))
(else (error "Unknown op: MAKE-FROM-REAL-IMAG" op))))

dispatch)

apply-generic⼿続きはジェネリック演算を引数に適⽤するものでしたが、こ
のデータに対応する apply-generic⼿続きは、単に演算の名前をデータオブジ
ェクトに渡して、オブジェクトに仕事をさせるだけになります。48

(define (apply-generic op arg) (arg op))

make-from-real-imagの返す値が⼿続き—内部⼿続き dispatch—であること
に注意してください。これは、apply-genericが演算の実⾏を要求したときに
呼び出される⼿続きです。
このプログラミングスタイルはメッセージパッシング (message passing)と

呼ばれています。この名前は、データオブジェクトは要求された演算の名前を
“メッセージ”として受け取る実体であるというイメージから来ています。メ
ッセージパッシングの例については、2.1.3節ですでに⾒ています。そこでは、
cons, car, cdrがデータオブジェクトを使わずに⼿続きだけを使って定義でき
るということを⾒てきました。ここでは、メッセージパッシングがただの数学
的トリックではなく、ジェネリック演算によってシステムを組み⽴てるうえで
の便利な技術だということを⽰しました。この章の残りでは、ジェネリックな
算術演算について検討するために、メッセージパッシングではなくデータ主導
プログラミングを使い続けることにします。第 3章ではメッセージパッシング

48この仕組みの制約のひとつは、⼀引数のジェネリック⼿続きしか使えないというこ
とです。

に戻り、それがシミュレーションプログラムを構築するうえでの強⼒なツール
になるということを⾒ていきます。

練習問題 2.75: コンストラクタ make-from-mag-angをメッセージ
パッシングスタイルで実装せよ。この⼿続きは、上で与えた make-
from-real-imagと似たようなものになるだろう。
練習問題 2.76: ジェネリック演算を使った⼤きなシステムが発展
するにつれ、新しいデータオブジェクトの型や新しい演算が必要
になることがある。三つの戦略 (明⽰的ディスパッチによるジェネ
リック演算、データ主導スタイル、メッセージパッシングスタイ
ル)それぞれについて、新しい型や新しい演算を追加するために必
要な変更を記述せよ。新しい型がよく追加されるシステムでは、ど
の組み⽴て⽅が最も適しているだろうか。新しい演算を追加する
システムでは、どれが最も適しているだろうか。

2.5 ジェネリック演算によるシステム
前の節では、データオブジェクトの表現が⼆通り以上あるシステムの設計⽅法
について⾒てきました。ポイントとなる考え⽅は、ジェネリックインターフェ
イスという⼿段によって、データ演算を指定するコードをいくつかの表現に結
びつけるというものでした。ここでは、同じ考え⽅を使って、いろいろな表現
に対してジェネリックな演算を定義するだけでなく、いろいろな種類の引数に
対してジェネリックな演算を定義するやり⽅について⾒ていきます。ここまで、
いくつもの算術演算パッケージに触れてきました。⾔語組み込みの基本算術演
算 (+, -, *, /)、2.1.1節の有理数演算 (add-rat, sub-rat, mul-rat, div-rat)、
それに2.4.3節で実装した複素数演算があります。今度はデータ主導テクニッ
クを使い、これまでに構築してきたすべての算術演算パッケージを合併した算
術演算パッケージを構築します。
図 2.23は、これから構築するシステムの構造を⽰しています。抽象化の壁

に注⽬してください。“数値”を使う⼈の視点からは、どんな種類の数が与えら
れても演算を⾏う addというたったひとつの⼿続きだけが⾒えています。add
はジェネリックインターフェイスの⼀部で、別々になっている通常の算術演算、
有理数の算術演算、複素数の算術演算のパッケージに、数値を使うプログラム
から統⼀的なアクセスができるようにします。個々の算術演算パッケージ (例
えば複素数パッケージ)は、それ⾃⾝、異なる表現 (例えば直交形式と極形式)

add-complex

mul-complex

ジェネリック算術演算パッケージ

通常の

リスト構造とマシンの基本算術演算

図 2.23: ジェネリック算術演算システム

のために設計された複数のパッケージを連結するジェネリック⼿続き (例え
ば add-complex)によってアクセスするようになっていることもあります。ま
た、システムの構造は加法的なので、それぞれの算術演算パッケージは別々に
設計して組み合わせてジェネリック算術演算システムを作れるようになってい
ます。

2.5.1 ジェネリック算術演算
ジェネリック算術演算を設計するというタスクは、ジェネリック複素数演算を
設計するのと似たようなものになります。例えば、通常の数値に対しては基本
⼿続きの⾜し算 +のようにふるまい、有理数に対しては add-ratのようにふる
まい、複素数に対しては add-complexのようにふるまうジェネリックな⾜し算
⼿続き addがほしいところです。addその他のジェネリック算術演算は、2.4.3
節で複素数のジェネリックなセレクタの実装に使ったのと同じ戦略によって実
装できます。それぞれの種類の数値にタイプタグをくっつけ、ジェネリック⼿
続きが引数のデータ型によって適切なパッケージにディスパッチを⾏うように
します。

ジェネリック算術演算⼿続きは、次のように定義します。
(define (add x y) (apply-generic 'add x y))
(define (sub x y) (apply-generic 'sub x y))
(define (mul x y) (apply-generic 'mul x y))
(define (div x y) (apply-generic 'div x y))

まず、通常の (ordinary)数値、つまり⾔語の基本数値を扱うパッケージを組み
込みます。通常の数値には scheme-numberという記号のタグをつけることに
します。このパッケージの算術演算は基本算術演算⼿続きです (つまり、タグ
なし数値を扱うのに新たに⼿続きを定義する必要はありません)。これらの演
算はそれぞれ⼆つの引数を取るので、テーブルには (scheme-number scheme-
number)というリストをキーとして組み込みます。
(define (install-scheme-number-package)

(define (tag x) (attach-tag 'scheme-number x))
(put 'add '(scheme-number scheme-number)

(lambda (x y) (tag (+ x y))))
(put 'sub '(scheme-number scheme-number)

(lambda (x y) (tag (- x y))))
(put 'mul '(scheme-number scheme-number)

(lambda (x y) (tag (* x y))))
(put 'div '(scheme-number scheme-number)

(lambda (x y) (tag (/ x y))))
(put 'make 'scheme-number (lambda (x) (tag x)))
'done)

scheme-number パッケージのユーザーは、次の⼿続きによって (タグつきの)
通常の数値を作成します。
(define (make-scheme-number n)

((get 'make 'scheme-number) n))

ジェネリック算術演算システムの枠組みがもうできているので、新しい種類の
数値を加えることは簡単にできます。以下は、有理数算術演算を⾏うパッケー
ジです。加法性のおかげで、2.1.1節の有理数のコードを修正なしでパッケージ
の内部⼿続きとして使えるというところに注⽬してください。
(define (install-rational-package)

;; 内部⼿続き

(define (numer x) (car x))
(define (denom x) (cdr x))
(define (make-rat n d)

(let ((g (gcd n d)))
(cons (/ n g) (/ d g))))

(define (add-rat x y)
(make-rat (+ (* (numer x) (denom y))

(* (numer y) (denom x)))
(* (denom x) (denom y))))

(define (sub-rat x y)
(make-rat (- (* (numer x) (denom y))

(* (numer y) (denom x)))
(* (denom x) (denom y))))

(define (mul-rat x y)
(make-rat (* (numer x) (numer y))

(* (denom x) (denom y))))
(define (div-rat x y)

(make-rat (* (numer x) (denom y))
(* (denom x) (numer y))))

;; システムのほかの部分とのインターフェイス
(define (tag x) (attach-tag 'rational x))
(put 'add '(rational rational)

(lambda (x y) (tag (add-rat x y))))
(put 'sub '(rational rational)

(lambda (x y) (tag (sub-rat x y))))
(put 'mul '(rational rational)

(lambda (x y) (tag (mul-rat x y))))
(put 'div '(rational rational)

(lambda (x y) (tag (div-rat x y))))
(put 'make 'rational

(lambda (n d) (tag (make-rat n d))))
'done)

(define (make-rational n d)
((get 'make 'rational) n d))

複素数を扱う同様のパッケージを、complex というタグを使って組み込みま
す。このパッケージを作る際に、直交形式と極形式のパッケージで定義した演

算 make-from-real-imagと make-from-mag-angをテーブルから取り出してい
ます。加法性のおかげで、2.4.1節の⼿続き add-complex, sub-complex, mul-
complex, div-complex が内部⼿続きとして使えます。
(define (install-complex-package)

;; 直交形式パッケージと極形式パッケージからインポートした⼿続き
(define (make-from-real-imag x y)

((get 'make-from-real-imag 'rectangular) x y))
(define (make-from-mag-ang r a)

((get 'make-from-mag-ang 'polar) r a))
;; 内部⼿続き
(define (add-complex z1 z2)

(make-from-real-imag (+ (real-part z1) (real-part z2))
(+ (imag-part z1) (imag-part z2))))

(define (sub-complex z1 z2)
(make-from-real-imag (- (real-part z1) (real-part z2))

(- (imag-part z1) (imag-part z2))))
(define (mul-complex z1 z2)

(make-from-mag-ang (* (magnitude z1) (magnitude z2))
(+ (angle z1) (angle z2))))

(define (div-complex z1 z2)
(make-from-mag-ang (/ (magnitude z1) (magnitude z2))

(- (angle z1) (angle z2))))
;; システムのほかの部分とのインターフェイス
(define (tag z) (attach-tag 'complex z))
(put 'add '(complex complex)

(lambda (z1 z2) (tag (add-complex z1 z2))))
(put 'sub '(complex complex)

(lambda (z1 z2) (tag (sub-complex z1 z2))))
(put 'mul '(complex complex)

(lambda (z1 z2) (tag (mul-complex z1 z2))))
(put 'div '(complex complex)

(lambda (z1 z2) (tag (div-complex z1 z2))))
(put 'make-from-real-imag 'complex

(lambda (x y) (tag (make-from-real-imag x y))))
(put 'make-from-mag-ang 'complex

(lambda (r a) (tag (make-from-mag-ang r a))))

3 4complex rectangular

図 2.24: 直交形式による 3 + 4iの表現

'done)

複素数パッケージの外部のプログラムは、複素数を構築するのに実部と虚部を
使うこともできますし、絶対値と偏⾓を使うこともできます。元の⼿続きは、
本来は直交形式パッケージと極形式パッケージの中で定義されたものですが、
そこから複素数パッケージにエクスポートされ、さらにそこから外の世界へと
エクスポートされているというところに注⽬してください。
(define (make-complex-from-real-imag x y)

((get 'make-from-real-imag 'complex) x y))
(define (make-complex-from-mag-ang r a)

((get 'make-from-mag-ang 'complex) r a))

ここでは、⼆つのレベルのタグシステムができています。3 + 4iのような典型
的な複素数は、図 2.24に⽰すように表現されることになります。外側のタグ
(complex)は、数値を複素数パッケージに送るために使われます。複素数パッケ
ージに⼊ると、次のタグ数値を直交形式パッケージに送るのに (rectangular)
タグが使われます。巨⼤で複雑なシステムでは、多くのレベルが、それぞれ次
のレベルとジェネリック演算という⼿段によって接続されるという形で存在す
るということがありえます。データオブジェクトが “下向きに”渡されるにつ
れ、適切なパッケージに送るために使われた外側のタグは (contentsの適⽤に
よって)はがされ、(もしあれば)次のレベルのタグが⾒えるようになり、さら
なるディスパッチに使われます。
上記のパッケージでは、add-rat, add-complexその他の算術演算⼿続きは、

最初に書いたときそのままの状態で使いました。しかし、これらの定義が異な
るインストール⼿続きの内部に⼊ると、互いに識別できる名前である必要はな
くなります。両⽅のパッケージで、単に add, sub, mul, divという名前をつけ
ていたとしても⼤丈夫です。

練習問題 2.77: Louis Reasoner は、式 (magnitude z) を評価し
ようとした。ここで、z は 図 2.24に⽰したオブジェクトである。
驚いたことに、5 という答えが返ってくるのではなく、apply-
genericからのエラーメッセージが返ってきて、型 (complex)に
は magnitude という演算を⾏う⼿続きがないという。彼がこの
対話を Alyssa P. Hackerに⾒せると、Alyssaは “複素数セレクタ
が polar と rectangular の数値にだけしか定義されていなくて、
complexの数値に対して定義されていないのが問題なのよ。動く
ようにするには complexパッケージに以下の式を追加すれば⼤丈
夫”と⾔う。
(put 'real-part '(complex) real-part)
(put 'imag-part '(complex) imag-part)
(put 'magnitude '(complex) magnitude)
(put 'angle '(complex) angle)

なぜこれが動くのか、詳しく説明せよ。例として、zが図 2.24に
⽰したオブジェクトであるとき、式 (magnitude z)を評価した際
に呼ばれるすべての⼿続きをトレースせよ。具体的には、apply-
genericは何回起動されるだろうか。それぞれの場合について、デ
ィスパッチ先の⼿続きは何になるだろうか。
練習問題 2.78: scheme-numberパッケージの内部⼿続きは、本質
的に基本⼿続き +, - その他に対する呼び出しでしかない。ここで
のタイプタグシステムでは、各データオブジェクトに型がくっつ
いていないといけないようになっていたので、直接この⾔語の基
本⼿続きを使うことはできなかった。しかし、実際のところ、Lisp
の実装はすべて型システムを持っていて、内部で使⽤している。
symbol?や number?のような基本述語は、データオブジェクトが
特定の型を持つか識別している。2.4.2節の type-tag, contents,
attach-tag の定義を変更し、ここでのジェネリックシステムが
Schemeの内部型システムを利⽤できるようにせよ。つまり、シス
テムの挙動はそのままで、普通の数値を carが scheme-numberと
いう記号であるペアとして表すのではなく、単に Schemeの数値
として表現されるようにせよ。
練習問題 2.79: ⼆つの数値の等価性をテストするジェネリックな
等価性述語 equ?を定義し、ジェネリック算術演算パッケージに組

み込め。この演算は、通常の数値、有理数、複素数に対して動作し
なければならない。
練習問題 2.80: 引数が 0であるかテストするジェネリックな述語
=zero?を定義し、ジェネリック算術演算パッケージに組み込め。
この演算は、通常の数値、有理数、複素数に対して動作しなければ
ならない。

2.5.2 異なる型のデータを組み合わせる
ここまで、統合算術演算システムの定義の仕⽅について⾒てきました。このシ
ステムは、通常の数値、複素数、有理数を含み、またこのほかどんな数値のタ
イプでも作って組み込むことができるものでした。しかし、無視してきた重要
な問題がひとつあります。ここまでで定義してきた演算は、異なるデータ型を
完全に独⽴したものとして扱うようになっています。つまり、例えば⼆つの通
常の数字を⾜すのと⼆つの複素数を⾜すのに別々のパッケージがあるというこ
とです。考えの対象外にしていたのは、型の境界を超えた演算、例えば複素数
と通常の数値との⾜し算のようなものについて考えることに意味があるという
事実です。私たちは、プログラムの部品同⼠の間に壁を作り、独⽴して開発し
たり理解したりできるようにすることに多⼤な労⼒をつぎ込んできました。こ
こでは、注意深くコントロールされたやり⽅でクロスタイプ (異なる型同⼠)の
演算を導⼊し、モジュール境界を⼤きく壊すことなくそれらの演算をサポート
できるようにします。
クロスタイプ演算を扱う⽅法のひとつは、可能な型の組み合わせの中で、

演算が有効であるものそれぞれに対して別々の⼿続きを設計するというもので
す。例えば、複素数パッケージを拡張し、複素数と実数の⾜し算を提供するよ
うにして、(complex scheme-number)というタグを使ってテーブルに組み込む
ということができます。49

;; 複素数パッケージに含める
(define (add-complex-to-schemenum z x)

(make-from-real-imag (+ (real-part z) x) (imag-part z)))
(put 'add '(complex scheme-number)

(lambda (z x) (tag (add-complex-to-schemenum z x))))

49このほかに、(scheme-number complex)という⼆つの型を扱うために、ほとんど同
じ⼿続きをもうひとつ⽤意する必要があります。

このテクニックはうまくいくのですが、⾯倒です。このようなシステムでは、
新しい型を導⼊するコストが、その型を扱う⼿続きのパッケージの構築だけで
は終わらず、クロスタイプ演算を実装する⼿続きを構築して組み込む分まで必
要になります。これはあっという間に、その型⾃⾝の演算を定義するのに必要
な分よりずっと多い量になってしまいます。この⼿法はまた、別々のパッケー
ジを加法的に組み合わせる能⼒を台無しにしてしまいます。少なくとも、個々
のパッケージを実装する⼈がほかのパッケージについてあまり考慮しないでい
いようにするという能⼒は台無しになってしまいます。例えば、上の例では、
複素数と通常の数値の混合演算を扱うのが複素数パッケージの責任だというこ
とは妥当に⾒えます。しかし、有理数と複素数を組み合わせるのは、複素数パ
ッケージでやってもいいし、有理数パッケージでやってもいいし、これら⼆つ
のパッケージから演算を抽出する何らかの第三のパッケージでやってもいいか
もしれません。複数のパッケージにわたる責任の分割について⼀貫したポリシ
ーを策定するということは、多くのパッケージと多くのクロスタイプ演算を伴
うシステムを設計する際には、どうしようもないほど⼤変なタスクになってし
まうでしょう。

強制型変換
まったく関係のない型同⼠のまったく関連のない演算という⼀般的な状況で
は、たとえ⾯倒でも、明⽰的なクロスタイプ演算を実装することぐらいしかで
きません。幸い普通は、型システムに隠れているかもしれない積み上げ式の構
造を利⽤して、もっとうまくやることができます。別々のデータ型が完全に独
⽴しているというわけではなく、ある型のオブジェクトをほかの型と⾒なす⽅
法があるということはよくあります。このプロセスは、強制型変換 (coercion)
と呼ばれます。例えば、通常の数値と複素数を算術演算によって組み合わせる
ことを求められたら、通常の数値を虚部がゼロの複素数と⾒なすことができま
す。こうすると、この問題を⼆つの複素数を組み合わせる問題に変換でき、複
素数パッケージによって普通のやり⽅で扱うことができるようになります。
⼀般に、ある型のオブジェクトを等価なほかの型のオブジェクトに変換す

る強制型変換⼿続きを設計することによって、この考え⽅を実装できます。以
下のものは典型的な強制型変換⼿続きで、与えられた通常の数値を、実部がそ
の数値で虚部がゼロの複素数に変換します。
(define (scheme-number->complex n)

(make-complex-from-real-imag (contents n) 0))

これらの強制型変換⼿続きは、特別な強制型変換テーブルに組み込むことにし
ます。⼆つの型の名前をキーとして使います。
(put-coercion 'scheme-number

'complex
scheme-number->complex)

(このテーブルを操作するための⼿続き put-coercionと get-coercionが存在
すると仮定しています)⼀般に、このテーブルにはいくつか空きができます。す
べての型の任意のデータオブジェクトをほかのすべての型に強制型変換すると
いうことは、⼀般的には可能でないからです。例えば、任意の複素数を実数に
強制型変換する⽅法というものはありません。そのため、⼀般的な complex-
>scheme-number⼿続きがテーブルに含まれるということはありません。
強制型変換テーブルが準備できたら、2.4.3節の apply-generic⼿続きに変

更を加えることによって、強制型変換を統⼀的に扱うことができるようになり
ます。演算を適⽤することを求められたら、最初はこれまでと同じように引数
の型に対して演算が定義されているかどうかチェックします。定義されていれ
ば、演算-型テーブルで⾒つかった⼿続きにディスパッチします。定義されてい
なければ、強制型変換を試みます。簡単にするために、ここでは引数が⼆つの
場合だけを考えます。50強制型変換テーブルを⾒て、⼀つ⽬の型のオブジェク
トが⼆つ⽬の型に強制型変換可能かどうか確認します。可能であれば、⼀つ⽬
の引数を強制型変換し、もう⼀度演算を試みます。もし⼀つ⽬の型のオブジェ
クトが⼆つ⽬の型に⼀般的に強制型変換できないのであれば、逆に⼆つ⽬の引
数を⼀つ⽬の引数の型に強制型変換できないか試してみます。最後に、どちら
の型についてももう⼀⽅の型に強制型変換する既知の⽅法がなければ、あきら
めます。⼿続きは以下のようになります。
(define (apply-generic op . args)

(let ((type-tags (map type-tag args)))
(let ((proc (get op type-tags)))

(if proc
(apply proc (map contents args))
(if (= (length args) 2)

(let ((type1 (car type-tags))
(type2 (cadr type-tags))
(a1 (car args))

50⼀般化については練習問題 2.82を参照。

(a2 (cadr args)))
(let ((t1->t2 (get-coercion type1 type2))

(t2->t1 (get-coercion type2 type1)))
(cond (t1->t2

(apply-generic op (t1->t2 a1) a2))
(t2->t1
(apply-generic op a1 (t2->t1 a2)))

(else (error "No method for these types"
(list op type-tags))))))

(error "No method for these types"
(list op type-tags)))))))

上でざっくり⾒たように、この強制型変換という構想は、明⽰的にクロスタイ
プ演算を定義するという⼿法に⽐べて多くのメリットがあります。型同⼠を関
連づけるために強制型変換⼿続きを書く (n個の型を持つシステムに対しては、
最⼤で n2個の⼿続き)ということはやはり必要ですが、書く必要があるのはそ
れぞれの型のペアに対して⼿続きひとつだけで、それぞれの型の集合とそれぞ
れのジェネリック演算に対して別々の⼿続きを書く必要はありません。51ここ
では、型同⼠の適切な変換は型そのものによって決まり、適⽤する演算には依
存しないという事実を利⽤しています。
⼀⽅、この強制型変換構想の持つ汎⽤性では⼗分でない応⽤もあるかもし

れません。組み合わせるオブジェクトのどちらももう⼀⽅の型に変換できない
という場合でも、両⽅のオブジェクトを第三の型に変換することで演算の実⾏
が可能かもしれません。そのような複雑な場合を扱いながらプログラムのモジ
ュール性を維持するためには、普通は型同⼠の関係のさらに深い構造を利⽤す
るシステムを構築することが必要になります。これについては次節で検討し
ます。

51うまくやると、普通は n2 よりも少ない数の強制型変換⼿続きで済ませることができ
ます。例えば、もし型 1から型 2への変換⽅法と型 2から型 3への変換⽅法がわかって
いれば、この知識を使って型 1から型 3に変換できます。このやり⽅を使うと、システ
ムに新しい型を追加する際に明⽰的に提供する必要のある強制型変換⼿続きの数を劇的
に減らすことができます。もしシステムを必要なだけ⾼度なものにする準備があるのな
ら、システムが型同⼠の関係の “グラフ”を検索して、明⽰的に提供された強制型変換⼿
続きから推論できる強制型変換⼿続きを⾃動的に⽣成させることもできます。

複素数

実数

有理数

整数

図 2.25: 型のタワー

型の階層
上で紹介した強制型変換構想は、型のペアの間に⾃然な関係があるということ
を頼りにしていました。しかし、異なる型の間の関係には、より “グローバル
な”構造があるということがよくあります。例えば、整数、有理数、実数、複素
数を扱うジェネリック算術演算システムを構築しているとしましょう。そのよ
うなシステムでは、整数を特別な種類の有理数と⾒なし、有理数を特別な種類
の実数と⾒なし、実数を特別な種類の複素数と⾒なすというのはとても⾃然で
す。ここで実際に扱っているのはいわゆる型の階層 (hierarchy of types)という
もので、その中では、例えば整数は有理数のサブタイプ (subtype)(つまり、有
理数に適⽤できる任意の演算は⾃動的に整数に適⽤できる)ということになり
ます。逆に、有理数は整数のスーパータイプ (supertype)と呼びます。ここで
扱っている特定の階層はとても単純なもので、それぞれの型は⾼々ひとつのス
ーパータイプと⾼々ひとつのサブタイプを持ちます。このような構造はタワー
(tower)と呼ばれます。図 2.25にこの構造を⽰します。
もし扱っているのがタワー構造であれば、階層に新しい型を追加するとい

う問題はとても簡単にできます。新しい型をすぐ上のスーパータイプに組み⼊
れる⽅法と、新しい型がすぐ下の型のどのようなスーパータイプであるかを記
述するだけで⼗分だからです。例えば、複素数と整数の⾜し算をしたい場合、
明⽰的に integer->complex という特別な強制型変換⼿続きを定義する必要は
ありません。その代わりに、整数の有理数への変換⽅法、有理数の実数への変
換⽅法、実数の複素数への変換⽅法を定義します。次に、システムがこれらの
ステップを通して整数を複素数に変換できるようにして、それから⼆つの複素
数を⾜し合わせます。

apply-generic⼿続きは、次のように設計し直すことができます。それぞ

れの型に対して、その型のオブジェクトをタワーの⼀階上に “上げる”raiseと
いう⼿続きを⽤意します。こうすると、システムが異なる型のオブジェクトの
演算を⾏うよう求められたとき、すべてのオブジェクトが塔の同じ階に揃うよ
うになるまで低い型を連続して上げていくということができるようになります
(練習問題 2.83と練習問題 2.84は、そのような戦略の実装の詳細に関係するも
のです)。
タワーの別の利点として、すべての型がスーパータイプに定義されたすべ

ての演算を “継承する”という概念を簡単に実装できるということがあります。
例えば、整数に対して実部を求める特別な⼿続きを提供していないとしても、
整数は複素数のサブタイプなので、整数に対しても real-partが定義されてい
ることが期待されます。タワーであれば、apply-genericを修正して、統⼀的
な⽅法でこれを実現できます。もし必要な演算が与えられたオブジェクトの型
に対して直接定義されていなければ、オブジェクトをそのスーパータイプに上
げて再試⾏します。こうやって、望む演算が実⾏可能になる階にたどり着くか、
てっぺんに当たるか (その場合はあきらめることになります)するまで、引数を
変換しながらタワーを登っていきます。
より⼀般的な階層に対してのタワーのもうひとつの利点は、データオブジ

ェクトを最も単純な表現に “下げる”ことが簡単にできるということがありま
す。例えば、2 + 3iと 4− 3iを⾜す場合、複素数 6 + 0iという形で答えを得る
よりも、整数の 6という答えを得るほうがいいでしょう。練習問題 2.85では、
そのようなレベル下げ演算について検討します (注意すべきところは、レベル
下げが可能な 6 + 0iのようなオブジェクトと、レベル下げが不可能な 6 + 2iの
ようなオブジェクトを⾒分ける⼀般的な⽅法が必要だというところです)。

階層の不適切さ
システムのデータ型が⾃然にタワーとして配置できる場合、ここまで⾒てきた
ように、異なる型同⼠のジェネリック演算の問題はとても単純になります。残
念ながら、普通はそうはいきません。図 2.26 はいろいろな型のより複雑な関係
を図⽰したものです。この例では、さまざまな型の幾何学図形同⼠の関係を⽰
しています。この図から、⼀般的にはひとつの型が⼆つ以上のサブタイプを持
つということが⾒て取れます。例えば、三⾓形と四⾓形はどちらも多⾓形のサ
ブタイプです。それに加えて、ひとつの型が⼆つ以上のスーパータイプを持つ
こともあります。例えば、直⾓⼆等辺三⾓形は⼆等辺三⾓形と⾒なすこともで
きますし、直⾓三⾓形と⾒なすこともできます。この複数スーパータイプ問題
は特に厄介です。ある型を階層の中で “上げる”ための唯⼀の⽅法というもの

がないということになるからです。あるオブジェクトにある演算を適⽤する際
に、“正しい”スーパータイプを⾒つけるために、apply-genericのような⼿続
きが型ネットワーク全体の中をかなり検索しないといけないことになるかもし
れません。⼀般的に、ひとつの型には複数のサブタイプがあるので、値を型階
層の中で “下げる”強制型変換にも同じような問題があります。巨⼤システム
を設計する際に、モジュール性を保ちながら相互に関連する多くの型を取り扱
うということは⾮常に難しく、現在多くの研究がなされている領域です。52

練習問題 2.81: Louis Reasonerは、引数の型がすでに同じであっ
ても、apply-genericは引数をお互いの型に強制型変換しようと
してもいいのではないかと気がついた。そのため、それぞれの型
の引数をそれ⾃⾝の型に強制型変換 (coerce)する⼿続きを強制型
変換テーブルに⼊れる必要があると彼は考えた。例えば、上に⽰
した scheme-number->complexという強制型変換に加え、彼は次
のことを⾏う。
(define (scheme-number->scheme-number n) n)
(define (complex->complex z) z)
(put-coercion 'scheme-number

'scheme-number
scheme-number->scheme-number)

(put-coercion 'complex 'complex complex->complex)

a. Louisの強制型変換⼿続きを組み込むと、もし scheme-number
型の⼆つの引数や complex 型の⼆つの引数とある演算に対

52この⽂は第⼀版にもあったものですが、12年前と同じように、現状に対しても当て
はまります。いろいろな型を持つ実体同⼠の関係 (哲学者が “オントロジー”と呼ぶもの)
を表現する実⽤的で汎⽤的な枠組みを開発するということは、どうしようもないほど難
しいように思えます。10年前に存在した混乱といま存在する混乱の主な違いは、今では
いろいろな不適切なオントロジーの理論が、同じように不適切なプログラミング⾔語の
⼭に組み⼊れられているということです。例えば、オブジェクト指向⾔語の複雑性—そ
れと、現代的なオブジェクト指向⾔語同⼠の微妙で混乱を招く相違点—の多くは、相互
に関連する型に対するジェネリック演算の扱いが中⼼となっています。第 3章では計算
オブジェクトについて検討しますが、これらの問題を完全に回避しています。オブジェ
クト指向⾔語に慣れた読者は、第 3章で局所状態についてかなり紙⾯を費やしているの
に、“クラス”や “継承”には⼀⾔も触れていないことに気づくと思います。実際のとこ
ろ、知識表現や⾃動推論の研究成果を利⽤することなしに、これらの問題をコンピュー
タ⾔語設計という観点のみから解決することはできないのではないかと私たちは疑って
います。

図 2.26: 幾何学図形の型の間の関係

して apply-genericが呼ばれ、その演算がテーブル内でそれ
らの型に対して⾒つからない場合は、何が起こるだろうか。
例えば、ジェネリックな指数関数演算を定義したとする。
(define (exp x y) (apply-generic 'exp x y))

そして、Scheme-numberパッケージに対して指数関数⼿続き
を追加し、ほかのパッケージには追加しないとする。
;; 以下は Scheme-numberパッケージに追加する
(put 'exp '(scheme-number scheme-number)

(lambda (x y) (tag (expt x y))))
; 基本⼿続き exptを使う

expを⼆つの複素数の引数で呼び出した場合、何が起こるだ
ろうか。

b. 同じ型の引数に対する強制型変換について⼿を加えないと
いけないとする Louis の考え⽅は正しいだろうか。それと

も、apply-genericはそのままの状態で正しく動作するだろ
うか。

c. apply-genericを修正し、⼆つの引数が同じ型であれば強制
型変換を試⾏しないようにせよ。

練習問題 2.82: apply-genericを⼀般化し、複数の引数⼀般につ
いて強制型変換を扱うようにするやり⽅を⽰せ。戦略のひとつと
しては、すべての引数を⼀つ⽬の引数の型に強制型変換すること
を試み、次に⼆つ⽬の引数の型に強制型変換することを試み、と
いうことを続けるというものだ。この戦略について (また、上で述
べた⼆引数バージョンについても)、それが⼗分に⼀般的でないよ
うな状況の例を挙げよ (ヒント：テーブルには適切な混合型演算
があり、それが試⾏されないという場合について考える)。
練習問題 2.83: 図 2.25に⽰した型のタワー (整数、有理数、実数、
複素数)を扱うジェネリック算術演算システムを設計していると
する。それぞれの型 (複素数を除く)に対し、その型のオブジェク
トをタワーの中で⼀階上げる⼿続きを設計せよ。それぞれの型 (複
素数を除く)に対して動作するジェネリックな raise演算を組み
込む⽅法を⽰せ。
練習問題 2.84:練習問題 2.83の raise演算を使って apply-generic
⼿続きを修正して、この節で検討した通り、連続して “上げる”と
いう⽅法によって引数が同じ型を持つよう強制型変換を⾏うよう
にせよ。⼆つの型のどちらがタワーの中で⾼い位置にあるかをテ
ストする⽅法を考える必要がある。システムのほかの部分と “互換
性がある”ようなやり⽅でこれを⾏い、タワーに新しい階を追加す
る際に問題を引き起こさないようにせよ。
練習問題 2.85: この節では、あるデータオブジェクトを型のタワ
ーの中で可能な限り下げていくことによって “単純化”するという
⽅法について触れた。練習問題 2.83で述べたタワーについて、こ
れを実⾏する⼿続き dropを設計せよ。ポイントは、オブジェクト
を下げることができるかどうかを何らかの汎⽤的なやり⽅で決め
るということにある。例えば、複素数 1.5 + 0iは realまで下げる
ことができ、複素数 1 + 0iは integerまで下げることができるが、
複素数 2+3iはまったく下げることができない。以下に、あるオブ
ジェクトを下げることができるかどうかを決める計画のひとつを

⽰す。まず、オブジェクトをタワーの中で⼀階下に “押す”ジェネ
リック演算 project (射影)を定義する。例えば、複素数の射影で
は虚部を捨てることになる。こうすると、ある数値を projectし
て、その結果を元の型に raiseしたときに、最初のものと等しい
何かになっていれば、その数値は下げることができるということ
になる。可能な限りオブジェクトを落とす⼿続き dropを書くこと
で、この考え⽅を実装するやり⽅を詳しく⽰せ。いろいろな射影
演算を設計し 53、projectをジェネリック演算としてシステムに
組み込むことが必要になる。また、練習問題 2.79で述べたジェネ
リック等価性述語を利⽤することも必要となる。最後に、dropを
使って練習問題 2.84の apply-genericを書き直し、解答を “単純
化”するようにせよ。
練習問題 2.86: 複素数の実部、虚部、絶対値、偏⾓に、通常の数値
や有理数や、これからシステムに追加するその他の数値を使える
ようにしたいとする。そのような複素数を使えるようにするため
に、システムにどのような変更を加える必要があるか説明し、そ
れを実装せよ。通常の数値と有理数に対してジェネリックに使え
る sineや cosineのような演算を定義する必要があるだろう。

2.5.3 例: 記号代数
記号代数式の操作は複雑な処理です。それについて検討することで、⼤規模シ
ステムの設計の際に起こる多くの困難な問題について知ることができます。代
数式は⼀般に、⽊構造の演算⼦を被演算⼦に適⽤する階層構造と⾒なすことが
できます。代数式を構築するには、定数や変数のような基本オブジェクトから
始めて、それらを加算や乗算のような代数演算⼦によって組み合わせます。ほ
かの⾔語の場合と同じように、複合オブジェクトを簡単に参照できるように抽
象化を⾏います。記号代数での典型的な抽象化としては、線形結合、多項式、
有理関数、三⾓関数などの考え⽅があります。これらは複合 “型”と⾒なすこと
ができ、そうすることで式の処理を指⽰することがやりやすくなります。例え
ば、次の式は、

x2 sin(y2 + 1) + x cos 2y + cos(y3 − 2y2)

整数を係数とする yの多項式の三⾓関数を係数とする xの多項式として記述す
ることができます。

53実数は、引数に最も近い整数を返す基本⼿続き roundを使って整数に射影できます。

ここでは完全な代数処理システムを開発しようとしているわけではありま
せん。そのようなシステムは、深い代数の知識と洗練されたアルゴリズムを具
現化した、⾮常に複雑なプログラムになります。ここでは、代数操作の単純で
ありながら重要な部分である多項式の算術演算について⾒ていきます。また、
そのようなシステムを設計する際にどのような決定をしなければいけないかと
いうことと、その取り組みに抽象データとジェネリック演算という考え⽅を適
⽤する⽅法について説明します。

多項式の算術演算
多項式の数値演算を⾏うシステムを設計するにあたっての最初の仕事は、多項
式とは何なのかを決めることです。多項式は通常、いくつかの変数 (多項式の不
定元 (indeterminate))に関して定義されます。簡単にするために、ここではひ
とつしか不定元を持たない多項式 (⼀元多項式 (univariate polynomial))に限定
することにします。54 多項式は項の和として定義し、その項は係数か、不定元
の冪乗か、係数と不定元の冪乗の積であるとします。係数は、多項式の不定元
と独⽴な代数式と定義します。例えば、

5x2 + 3x+ 7

は単純な xの多項式で、
(y2 + 1)x3 + (2y)x+ 1

は係数が yの多項式である xの多項式です。
ここまでで、もういくつかの厄介な問題を避けて通ってきています。上の

⼀番⽬の多項式は、多項式 5y2 + 3y + 7と同じものでしょうか、それとも異な
るものでしょうか。妥当な答えは、“多項式を純粋に数学の関数であると考え
れば、同じものである。しかし、多項式を構⽂形式であると考えれば、異なる
ものである”となるでしょう。また、⼆番⽬の多項式は、xの多項式を係数と
する yの多項式と代数学的に等価です。私たちのシステムはこのことを認識す
るべきでしょうか。しかも、多項式を表現する⽅法はほかにもあります—例え
ば、因数の積、(⼀元多項式の場合)根の集合、特定の点の集合における多項式
の値のリストなどによって表すこともできます。55 この代数操作システムで

54⼀⽅で、多項式の係数⾃⾝が別の変数に関する多項式であることは許すことにしま
す。こうすると、完全な多元システムと本質的に同じ表現⼒を持つことになりますが、こ
の先で述べるように、強制型変換の問題が出てきます。

55⼀元多項式について、与えられた点の集合における多項式の値を使うということが
特によい表現となることもあります。こうすると、多項式の算術演算は⾮常に単純にな

は、“多項式”というのは特定の構⽂形式であって、その奥にある数学的意味で
はないと決めることで、これらの問題を解決できます。
次に、多項式の算術演算をどのようなものにするか考える必要があります。

この単純なシステムでは、加算と乗算だけ考えることにします。さらに、組み
合わせる⼆つの多項式は同じ不定元を持っていなければならないということに
します。
このシステムの設計にあたっては、おなじみのデータ抽象化の規律に従い

ます。多項式はpolyというデータ構造を使って表現し、polyはひとつの変数と項
の集まりからなります。polyからそれらの部品を抽出するセレクタ variable,
term-listと、与えられた変数と項のリストから polyを組み⽴てるコンスト
ラクタ make-polyがあると想定します。変数はただの記号なので、変数の⽐較
には 2.3.2節の same-variable?⼿続きが使えます。以下の⼿続きは、polyの
加算と乗算を定義するものです。
(define (add-poly p1 p2)

(if (same-variable? (variable p1) (variable p2))
(make-poly (variable p1)

(add-terms (term-list p1) (term-list p2)))
(error "Polys not in same var: ADD-POLY" (list p1 p2))))

(define (mul-poly p1 p2)
(if (same-variable? (variable p1) (variable p2))

(make-poly (variable p1)
(mul-terms (term-list p1) (term-list p2)))

(error "Polys not in same var: MUL-POLY" (list p1 p2))))

多項式を私たちのジェネリック算術演算システムに組み⼊れるためには、タイ
プタグをつける必要があります。タグとしては polynomialを使うことにして、
タグつき多項式に対する適切な演算を演算テーブルに組み込みます。2.5.1節同
様、コードはすべて多項式パッケージの組み込み⼿続きの中に埋め込むことに
します。
(define (install-polynomial-package)

;; 内部⼿続き
;; polyの表現

ります。例えば、この⽅式で表現された⼆つの多項式の和を求めることは、対応する点
における多項式の値を⾜し合わせるだけでできます。より⾒慣れた表現に変形するには、
n + 1個の点における多項式の値から n次の多項式の係数を復元するラグランジュの補
間公式が使えます。

(define (make-poly variable term-list) (cons variable term-list))
(define (variable p) (car p))
(define (term-list p) (cdr p))
⟨2.3.2節の same-variable?と variable?⼿続き⟩
;; 項と項リストの表現
⟨下記の adjoin-term . . . coeff⼿続き⟩
(define (add-poly p1 p2) . . .)
⟨add-polyが使う⼿続き⟩
(define (mul-poly p1 p2) . . .)
⟨mul-polyが使う⼿続き⟩
;; システムのほかの部分とのインターフェイス
(define (tag p) (attach-tag 'polynomial p))
(put 'add '(polynomial polynomial)

(lambda (p1 p2) (tag (add-poly p1 p2))))
(put 'mul '(polynomial polynomial)

(lambda (p1 p2) (tag (mul-poly p1 p2))))
(put 'make 'polynomial

(lambda (var terms) (tag (make-poly var terms))))
'done)

多項式の加算は項ごとに実⾏します。同じ次数の項 (つまり、同じ指数を持つ
不定元) 同⼠を組み合わせる必要があります。これは、⾜す項と⾜される項の
係数の合計を係数とする、同じ次数を持つ新しい項を作ることで⾏えます。⽚
⽅にだけ項があって、もう⼀⽅のほうに同じ次数の項がない場合は、構築中の
和の多項式にその項をそのまま追加します。
項リストを操作するために、空の項リストを返す the-empty-termlistと

いうコンストラクタと、項リストに新しい項を追加する adjoin-termというコ
ンストラクタがあると想定します。さらに、与えられた項リストが空かどうか
調べる empty-termlist?という述語と、項リストから最⼤次数の項を取り出す
セレクタ first-term、最⼤次数の項以外のすべての項を返すセレクタ rest-
termsもあると想定します。項を操作するために、与えられた次数と係数から
項を構築するコンストラクタ make-termと、項の次数と係数をそれぞれ返すセ
レクタ orderと coeffがあると想定します。これらの演算によって、項と項リ
ストがデータ抽象として捉えられるようになり、具体的な表現は別に考えるこ
とができます。

以下は、⼆つの多項式の和となる項リストを構築する⼿続きです。56

(define (add-terms L1 L2)
(cond ((empty-termlist? L1) L2)

((empty-termlist? L2) L1)
(else
(let ((t1 (first-term L1))

(t2 (first-term L2)))
(cond ((> (order t1) (order t2))

(adjoin-term
t1 (add-terms (rest-terms L1) L2)))

((< (order t1) (order t2))
(adjoin-term
t2 (add-terms L1 (rest-terms L2))))

(else
(adjoin-term
(make-term (order t1)

(add (coeff t1) (coeff t2)))
(add-terms (rest-terms L1)

(rest-terms L2)))))))))

ここでの最も重要なポイントは、項の係数同⼠を⾜し合わせるのにジェネリッ
ク加算⼿続き addを使っているというところです。以下で⾒ていくように、こ
のことは強⼒な結果をもたらします。
⼆つの項リストを乗算するには、まず与えられた項と与えられた項リスト

のすべての項と乗算する mul-term-by-all-termsを繰り返し使って、最初の
リストの各項をもう⼀⽅のリストのすべての項と乗算します。次に、返される
項リストの列 (⼀つ⽬のリストの項それぞれに対してひとつずつ)を和として
集積します。⼆つの項の乗算では、各乗数の次数の和を次数とし、各乗数の係
数の積を係数とする項を作ります。
(define (mul-terms L1 L2)

(if (empty-termlist? L1)
(the-empty-termlist)
(add-terms (mul-term-by-all-terms (first-term L1) L2)

56この演算は、練習問題 2.62で開発した union-set演算にとても似ています。実際、
不定元の指数の順に並べた集合として多項式の項を考えると、和の項リストを⽣成する
プログラムは union-setとほとんど同じものになります。

(mul-terms (rest-terms L1) L2))))
(define (mul-term-by-all-terms t1 L)

(if (empty-termlist? L)
(the-empty-termlist)
(let ((t2 (first-term L)))

(adjoin-term
(make-term (+ (order t1) (order t2))

(mul (coeff t1) (coeff t2)))
(mul-term-by-all-terms t1 (rest-terms L))))))

多項式の加算と乗算は本当にこれがすべてです。項の演算にはジェネリック⼿
続き addと mulを使っているため、ジェネリック数値演算パッケージにとって
既知のどんな型の係数でも、⾃動的に多項式パッケージが扱えるようになって
いるというところに注意してください。2.5.2節で考察したような強制型変換
メカニズムを組み込めば、以下のような係数の型が異なる多項式に対する演算
も⾃動的に扱えるようになります。

[3x2 + (2 + 3i)x+ 7] ·
[
x4 +

2

3
x2 + (5 + 3i)

]
.

add-poly, mul-polyという多項式の加算と乗算の⼿続きを polynomial型
の addと mulとしてジェネリック算術演算システムに組み込んだので、私たち
のシステムは⾃動的に次のような多項式演算を扱えるようになっています。[

(y + 1)x2 + (y2 + 1)x+ (y − 1)
]
·
[
(y − 2)x+ (y3 + 7)

]
.

その理由は、システムが係数を組み合わせようとするとき、addと mulを通し
てディスパッチを⾏うためです。係数はそれ⾃⾝ (yの)多項式なので、これら
は add-polyと mul-polyを使って組み合わされることになります。結果とし
て、これは “データ主導再帰”のようなものになります。例えば、mul-poly を
呼び出すと、係数同⼠をかけ算するために mul-polyを再帰的に呼び出す結果
になるということです。もし係数の係数がそれ⾃⾝多項式なら (この⽅法は三
変数の多項式を表現するのに使えます)、データ主導の仕組みによってシステ
ムがもう⼀レベルの再帰を⾏うことが保証されます。データ構造によって規定
されるレベルの数だけ、この再帰は続きます。57

57この仕組みが完全にスムーズに動くようにするには、ジェネリック算術演算システ
ムに “数値”を多項式に強制型変換する能⼒を追加しなければなりません。そのために

項リストの表現
最後に、項リストのよい表現を実装するという仕事に取りかからなければなり
ません。項リストというのは、実質的には、項の次数をキーとする係数の集合
です。つまり、2.3.3節で検討したようなどんな集合の表現⼿法でも、このタス
クに適⽤できます。⼀⽅、⼿続き add-termsと mul-termsは、常に⾼い次数か
ら低い次数という⽅向で、項リストに順次アクセスします。このため、何らか
の順序つきリスト表現を使うことにします。
項リストを表現するリストはどのように構造化するべきでしょうか。考慮

するポイントのひとつは、操作対象となる多項式の “濃度”です。多項式は、ほ
とんどの次数の項が 0でない係数を持つ場合、密 (dense)と呼ばれます。もし
0の項が多ければ、疎 (sparse)と呼ばれます。例えば、

A : x5 + 2x4 + 3x2 − 2x− 5

は密多項式です。
B : x100 + 2x2 + 1

は疎です。
密な多項式の項リストは、係数のリストとして表現するのが最も効率的で

す。例えば上の Aは、(1 2 0 3 -2 -5)という形でうまく表現できます。この
表現では、その項の係数から始まるサブリストの⻑さから 1を引いたものにな
ります。58 しかしこれは、Bのような疎な多項式の表現としてはひどいものに
なります。ほんの少しの孤⽴した⾮ゼロ項によって区切られた、巨⼤なゼロの
リストになってしまうでしょう。疎な多項式のもうちょっとまともな表現とし
ては、⾮ゼロ項のリストというものがあります。それぞれの項は、項の次数と
その次数に対する係数を含むリストになります。その計画を使う場合、多項式
B は ((100 1) (2 2) (0 1))という形で効率的に表現されることになります。
ほとんどの多項式操作は疎な多項式に対して実⾏されるので、こちらの⼿法を
は、数値を 0次の多項式と⾒なし、係数がその数値であるとします。これは、次のよう
な演算を⾏う場合に必要になります。

[x2 + (y + 1)x+ 5] + [x2 + 2x+ 1],

この例では、係数 y + 1と係数 2を⾜すことが求められます。
58これらの多項式の例では、練習問題 2.78で提案したような型メカニズムを使ってジ
ェネリック算術演算システムを実装していると想定しています。つまり、通常の数値の
係数は、carが記号 scheme-numberであるペアとしてではなく、数値そのものによって
表現されているということです。

使うことにします。項リストは⾼次から低次の順に並べられた項のリストとし
て表現されると想定します。このことを決めれば、セレクタとコンストラクタ
は素直に実装できます。59

(define (adjoin-term term term-list)
(if (=zero? (coeff term))

term-list
(cons term term-list)))

(define (the-empty-termlist) '())
(define (first-term term-list) (car term-list))
(define (rest-terms term-list) (cdr term-list))
(define (empty-termlist? term-list) (null? term-list))
(define (make-term order coeff) (list order coeff))
(define (order term) (car term))
(define (coeff term) (cadr term))

=zero?は練習問題 2.80で定義したものです (下の練習問題 2.87も参照)。
多項式パッケージのユーザーは、(タグつきの)多項式を次の⼿続きによっ

て作成します。
(define (make-polynomial var terms)

((get 'make 'polynomial) var terms))

練習問題 2.87: 多項式に対する =zero?をジェネリック数値演算パ
ッケージに組み込め。こうすることで、係数がそれ⾃⾝多項式と
なっている多項式に対して adjoin-termが動作するようになる。
練習問題 2.88: 多項式システムを拡張し、多項式の減算ができる
ようにせよ (ヒント：ジェネリックな符号反転演算を定義すると
いいかもしれない)。
練習問題 2.89: 密な多項式に適していると上で述べた項リスト表
現を実装する⼿続きを定義せよ。

59項リストは順序つきだと想定していますが、adjoin-termの実装は、新しい項を既
存の項リストに単純に consするようになっています。adjoin-termを使う⼿続き (add-
termsなど)がこの⼿続きを呼ぶ際に、リスト内のものよりも⾼次の項を引数として呼
び出しを⾏うということを保証するなら、このままでも問題はありません。そのような
保証をしたくなければ、順序つきリストとしての集合の表現 (練習問題 2.61)に対する
adjoin-setに近い形で adjoin-termを実装することもできます。

練習問題 2.90: 疎な多項式と密な多項式の両⽅に対して効率的な
多項式システムを作りたいとする。そのためのひとつの⽅法は、シ
ステム内でどちらの項リスト表現も使えるようにするということ
である。この状況は2.4節の複素数の例と似ている。そのときは直
交形式と極形式のどちらの表現も使えるようにした。これを⾏う
ためには、異なる型の項リストを識別し、項リストに対する演算
をジェネリックにしなければならない。この⼀般化を実装するた
めに、多項式システムを再設計せよ。これは局所的な変更ではな
く、かなりの作業が必要になる。
練習問題 2.91: 1変数多項式は、別の 1変数多項式によって割るこ
とができる。商と剰余は多項式となる。例えば、次のようになる

x5 − 1

x2 − 1
= x3 + x, remainder x− 1.

除算は筆算⽅式で実⾏できる。まず、被除数の最⾼次の項を除数
の最⾼次の項で割る。その答えが商の第⼀項となる。次に、答えに
除数をかけ、被除数からそれを引く。このようにして、再帰的に引
き算の結果を除数で割り算するということを繰り返して残りの答
えを求める。除数の次数が被除数の次数を上回ったら終わりとし
て、被除数を剰余とする。また、もし被除数がゼロになった場合に
は、商と剰余の両⽅をゼロとして返す。
div-poly⼿続きは、add-polyや mul-polyをベースに設計できる。
この⼿続きは、⼆つの多項式が同じ変数を持つかどうかチェック
する。同じ変数を持つなら、div-polyは変数を取り去って問題を
div-termsに渡す。div-termsは項リストに対する除算を実⾏す
る。最後に、div-polyは div-termsの結果に変数をくっつけ直す。
div-termsは、除算の商と剰余の両⽅を求めるようにするのが便
利だ。div-termsは項リスト⼆つを引数として取り、商となる項リ
ストと剰余となる項リストからなるリストを返す。
次の空欄を埋め、div-termsの定義を完成させよ。これを使って、
⼆つの多項式を引数として取り、商と剰余の polyのリストを返す
div-polyを実装せよ。
(define (div-terms L1 L2)

(if (empty-termlist? L1)
(list (the-empty-termlist) (the-empty-termlist))
(let ((t1 (first-term L1))

(t2 (first-term L2)))
(if (> (order t2) (order t1))

(list (the-empty-termlist) L1)
(let ((new-c (div (coeff t1) (coeff t2)))

(new-o (- (order t1) (order t2))))
(let ((rest-of-result

⟨再帰的に残りを計算する⟩
))

⟨完全な結果を作る⟩
))))))

記号代数の型の階層
私たちの多項式システムは、ある型 (ここでは多項式)のオブジェクトが、実際
には多くの異なる型のオブジェクトを部品として持つ複雑なオブジェクトだと
いうことがありうることを⽰しています。このことは、ジェネリック演算を定
義するうえで特に難しい問題にはなりません。必要な複合型の部品の操作を⾏
うための適切なジェネリック演算を組み込むだけで⼤丈夫です。実際、多項式
は “再帰的データ抽象化”の⼀種となっているということを⾒てきました。こ
れは、多項式の部品がそれ⾃⾝多項式でありうるということです。ジェネリッ
ク演算とデータ主導プログラミングスタイルのおかげで、それほど苦労せずに
この複雑な問題を扱うことができています。
⼀⽅、多項式代数は、データ型を⾃然にタワー型に配置できないシステム

です。例えば、このシステムでは係数が yの多項式である xの多項式というも
のが作れます。また、係数が xの多項式である yの多項式を作ることもできま
す。これらの型は、⾃然に考えると、⽚⽅がもう⼀⽅の “上”であるとはどうや
っても⾔えません。しかし、それぞれの集合の要素を⾜し合わせることが必要
になるということはよくあります。ひとつの可能性としては、⽚⽅の多項式を
展開したり再配置したりして、もう⼀⽅の多項式の型に変換し、両⽅の多項式
が同じ主変数を持つようにするというものです。変数に順番をつけて、どんな
多項式でも、⾼優先度の変数が主変数となり低優先度の変数が係数に埋め込ま
れるような “標準形”に常に変換することにすれば、タワー的な構造を押しつ
けることはできます。この戦略はそれなりにうまくいくのですが、変換の過程
で多項式を不必要に展開して、読みにくくなったり、場合によっては⾮効率的
になってしまったりします。このタワー戦略というのは、確実にこの領域に対
して⾃然なものとは⾔えません。ユーザーが古い型をいろいろなやり⽅ (三⾓
関数、冪級数、積分など)によって組み合わせて動的に新しい型を発明できる

ようなどのような領域に対しても、同じことが⾔えます。
強制型変換の制御は⼤規模な代数処理システムを設計するうえで深刻な問

題になります。これは驚くには当たりません。そのようなシステムでは、複雑
性の多くはいろいろな型同⼠の関係に関わるところにあります。実際のところ、
強制型変換についてはまだ完全に理解できているとは⾔えないところがありま
す。それどころか、データ型という概念についても、完全にわかっているとは
⾔えません。それでも、私たちの知識は、⼤規模なシステムの設計を⽀える強
⼒な構造化とモジュール化の原則を与えてくれます。

練習問題 2.92: 変数に順番をつけることによって、異なる変数の
多項式同⼠の加算と乗算が動作するよう多項式パッケージを拡張
せよ。(これは簡単ではない!)

拡張問題: 有理関数
ジェネリック算術演算システムを拡張し、有理関数 (rational function)を含む
ようにすることもできます。これは、次のように分⼦と分⺟が多項式の “分数”
です。

x+ 1

x3 − 1
.

システムは、有理関数の加算、減算、乗算、除算ができなければなりません。ま
た、以下のような計算も実⾏できる必要があります。

x+ 1

x3 − 1
+

x

x2 − 1
=
x3 + 2x2 + 3x+ 1

x4 + x3 − x− 1
.

(ここでは、加算結果は共通因数を取り除いて簡約されています。通常の “たす
き掛け”であれば、分⼦が 4次多項式で分⺟が 5次多項式分の分数になってい
たところです。)
有理数算術演算パッケージを修正し、ジェネリック演算を使えるようにす

れば、分数を既約にするという問題を除いて、望むことができるようになり
ます。

練習問題 2.93: 有理数算術演算パッケージを修正し、ジェネリッ
ク演算を使うようにせよ。しかし、make-ratは分数を既約にしよ
うと試みないように変更せよ。make-rationalを⼆つの多項式に
対して呼び出し有理関数を作ることで、システムをテストせよ。

(define p1 (make-polynomial 'x '((2 1)(0 1))))
(define p2 (make-polynomial 'x '((3 1)(0 1))))
(define rf (make-rational p2 p1))

次に、addを使って rfにそれ⾃⾝を⾜せ。この加算⼿続きが分数
を既約にしないことが確認できるだろう。

多項式の分数を既約にするのは、整数のところで使ったのと同じ考え⽅で⾏う
ことができます。make-ratを修正して、分⼦と分⺟の両⽅を、それらの最⼤公
約数で割るようにするということです。“最⼤公約数”という概念は、多項式に
ついても考えることができます。実際に、⼆つの多項式の gcd(最⼤公約数)は、
整数に対するユークリッドのアルゴリズムと本質的に同じものを使って求める
ことができます。60整数版は次のようなものでした。
(define (gcd a b)

(if (= b 0)
a
(gcd b (remainder a b))))

これを使って、⾃明な修正を加えて項リストに対して動く gcd演算を定義でき
ます。
(define (gcd-terms a b)

(if (empty-termlist? b)
a
(gcd-terms b (remainder-terms a b))))

ここでの remainder-terms は、練習問題 2.91で実装した項リスト除算演算
div-termsが返すリストから、剰余の部分を取り出すものです。

60ユークリッドのアルゴリズムが多項式に対して動作するということは、代数では、多
項式が ユークリッド環 (Euclidean ring)と呼ばれる代数領域の⼀種を形成するというよ
うに形式化されます。ユークリッド環とは、加算、減算、交換可能な乗算を認め、環の各
元 xに対して、ある性質を持った “測度”m(x)という正の整数を割り当てる⽅法を持つ
領域です。その性質とは、任意の⾮ゼロな xと yに対し m(xy) ≥ m(x)となり、任意の
xと yに対し、y = qx+ rで r = 0またはm(r) < m(x)となる qが存在するというもの
です。
抽象化の視点からこれがユークリッドのアルゴリズムがうまく⾏くのに必要な条件で

す。整数の定義域に対して、整数の⼤きさmはその整数の絶対値です。多項式の定義域
においては多項式の⼤きさはその次数です。

練習問題 2.94: div-termsを使って⼿続き remainder-termsを実
装し、それを使って gcd-termsを上記のように定義せよ。次に、⼆
つの多項式に対して多項式 gcdを求める⼿続き gcd-polyを書け
(この⼿続きは、⼆つの多項式が同じ変数のものでなければエラー
を出すようにする)。システムに greatest-common-divisor とい
うジェネリック演算を組み込み、多項式は gcd-polyによって簡約
し、通常の数値は通常の gcdによって簡約するようにせよ。テス
トとして以下を実⾏せよ。
(define p1 (make-polynomial

'x '((4 1) (3 -1) (2 -2) (1 2))))
(define p2 (make-polynomial 'x '((3 1) (1 -1))))
(greatest-common-divisor p1 p2)

その結果を⼿計算によって確認せよ。
練習問題 2.95: 以下の多項式 P1, P2, P3 を定義せよ。

P1 : x2 − 2x+ 1,

P2 : 11x2 + 7,

P3 : 13x+ 5.

次に、P1 と P2 の積 Q1、P1 と P3 の積 Q2 を定義し、greatest-
common-divisor(練習問題 2.94)を使って Q1 と Q2 の gcdを求め
よ。答えが P1 と同じにならないことに注意せよ。この例では計算
に⾮整数演算の演算が絡んでくるので、gcdアルゴリズムが困難
になっている。61 何が起こっているか理解するために、gcdの計
算時に gcd-termsをトレースせよ。または、この除算を⼿計算で
実⾏せよ。

練習問題 2.95で出てきた問題は、以下のように gcdアルゴリズムを修正する
と解決できます (これは、実際には整数係数の多項式の場合にしか動作しませ
ん)。gcd 計算で多項式の除算を⾏う前にはいつも、除算過程で分数が現れない
ことを保証するように選んだ整数定数を被除数にかけるようにします。こうす
ると、答えは実際の gcdと整数定数倍だけ違うものになりますが、このことは

61mit Schemeのような実装では、この計算は確かに Q1 と Q2 の約数となる多項式を
⽣成しますが、係数は有理数になります。ほかの多くの Schemeシステムでは、整数の
除算結果は有限精度の⼩数となることがあるので、正しい約数が得られないことがあり
ます。

有理関数を既約にする場合には問題になりません。gcdは分⼦と分⺟の両⽅を
割るのに使うので、整数定数倍は相殺されます。
より正確には、P とQが多項式であるとき、O1を P の次数 (つまり、P の

最⼤項の次数)とし、O2 を Qの次数とします。cを Qの先頭の係数とします。
すると、P に整数化因⼦ (integerizing factor)c1+O1−O2 をかけると、答えの多
項式を div-termsアルゴリズムを使って分数を出さずに Qで割ることができ
るということが⽰せます。被除数にこの定数をかけてから割るという操作は、
P の Qによる 擬除算 (pseudodivision)と呼ばれることもあります。除算の剰
余は 擬剰余 (pseudoremainder)と呼ばれます。

練習問題 2.96:

a. ⼿続き pseudoremainder-termsを実装せよ。これは remainder-
termsと同じようなものになるが、div-termsを呼ぶ前に、上
で述べた整数化因数を被除数にかけるというところが違う。
gcd-terms を修正して pseudoremainder-terms を使うよう
にし、greatest-common-divisorが練習問題 2.95の例に対し
て整数係数の答えを返すようになっていることを確認せよ。

b. これで gcdは整数係数になる。しかし、係数は P1 の係数よ
りも⼤きくなる。gcd-termsを修正し、答えの係数すべての
(整数の)最⼤公約数で係数を割ることによって答えの係数か
ら共通因数を取り除くようにせよ。

まとめると、有理関数を既約にする⽅法は以下のようになります。
• 練習問題 2.96版の gcd-termsを使って、分⼦と分⺟の gcdを求める。
• gcdを求めたら、分⼦と分⺟を gcdで割る前に、それらに同じ整数化因
⼦をかけて、gcdによる割り算で⾮整数の係数が出ないようにする。整
数化因⼦としては、gcdの最初の係数を 1 + O1 − O2 乗したものが使え
る。ここで、O2 は gcdの次数、O1 を分⼦と分⺟の⼤きいほうの次数と
する。こうすると、分⼦と分⺟を gcdで割っても分数が出てこないとい
うことが保証できる。

• この演算の結果は、整数係数を持つ分⼦と分⺟になる。いろいろ整数化
因⼦をかけているため、係数は普通とても⼤きなものになっている。そ
のため、最終ステップとして、分⼦と分⺟の係数すべてに対する (整数
の)最⼤公約数を求め、それで割ることによって、冗⻑な因数を取り除く。
練習問題 2.97:

a このアルゴリズムを reduce-termsという⼿続きとして実装
せよ。⼿続きは nと dという⼆つの項リストを引数として取
り、上で説明したアルゴリズムによって n と d を既約にし
た nnと ddのリストを返す。また、add-poly と同様の⼿続
き reduce-polyを実装せよ。これは⼆つの polyが同じ変数
を持つかどうかチェックし、もし同じであれば変数を取り去
って問題を reduce-termsに渡し、reduce-termsに返された
⼆つの項リストにその変数をつけ直す。

b reduce-termsと似たような⼿続きとして、元々の make-rat
が整数に対して⾏っていたことと同じことを実⾏するものを
次のように定義せよ。
(define (reduce-integers n d)

(let ((g (gcd n d)))
(list (/ n g) (/ d g))))

そして、reduceというジェネリック演算を定義せよ。これ
は、apply-genericを呼んで、reduce-poly(polynomial型引
数に対して) または reduce-integers(scheme-number 型引
数に対して)にディスパッチする。こうすると、make-ratが
分⼦と分⺟を組み合わせて有理数を作る前に reduceを呼ぶ
ようにすることで、有理数算術演算パッケージが有理数を既
約にするようにできる。これで、このシステムは整数と多項
式のどちらの有理式も扱えるようになった。プログラムをテ
ストするために、この拡張練習問題の最初に出てきた例を試
してみよ。
(define p1 (make-polynomial 'x '((1 1) (0 1))))
(define p2 (make-polynomial 'x '((3 1) (0 -1))))
(define p3 (make-polynomial 'x '((1 1))))
(define p4 (make-polynomial 'x '((2 1) (0 -1))))
(define rf1 (make-rational p1 p2))
(define rf2 (make-rational p3 p4))
(add rf1 rf2)

正しく既約になっている正しい答えが返ってくるかどうかを
確認せよ。

gcdの計算は、有理関数の演算を⾏うどんなシステムでも、中⼼的な存在にな
ります。上で使ったアルゴリズムは、数学的には素直なのですが、⾮常に遅い
ものです。遅さの原因は、部分的には除算回数の多さにあり、部分的には擬除
算によって⽣成される中間係数が巨⼤になるというところにあります。多項式
の gcdを求めるよりよいアルゴリズムをどう設計するかということは、代数処
理システム開発において活発な領域のひとつです。62

62多項式 gcd計算の⾮常に効率的でエレガントな⼿法が Richard Zippel (1979)によっ
て発⾒されました。この⼿法は、第 1章で検討した⾼速な素数性テストと同じく、確率
的アルゴリズムです。Zippelの本 (Zippel 1993)では、この⼿法のほかにも、多項式の
gcdを求めるほかのいくつかの⽅法について解説しています。

3
モジュール性、オブジェクト、状態

Mεταβάλλον ὰναπαύεται
(変化していながら、静⽌している)
—Heraclitus

Plus ça change, plus c’est la même chose.
(変化すればするほど、同じものとなる)
—Alphonse Karr

ここまでの章では、プログラムを作る基本的な要素を紹介してきました。基本
⼿続きや基本データを組み合わせて合成物を作る⽅法について概観し、⼤きな
システムの複雑性に対処するには抽象化が重要だということを学びました。し
かし、これらの道具はプログラムを設計するのに⼗分ではありません。プログ
ラムを効率的に統合していくためには、プログラムの全体的な設計計画を⽴て
るうえで助けになってくれる組織化原則が必要です。とりわけ、⼤きなシステ
ムをモジュール式 (modular)に構成する、つまり、システムをまとまりのある
部品に “⾃然に”分割し、別々に開発、保守をすることを可能にするための戦略
が必要となります。

233



物理システムをモデル化するプログラムの構成に特に適した強⼒な設計戦
略のひとつとして、モデル化対象のシステムの構造をもとにプログラムの構造
を決めるというものがあります。システムのそれぞれのオブジェクトに対して、
対応する計算オブジェクトを構築するというものです。この戦略をとることの
狙いは、新しいオブジェクトやアクションを導⼊するためにモデルを拡張する
際に、プログラムの戦略はそのままで、それらのオブジェクトやアクションの
記号的な対応物を追加するだけですむようにすることです。システムの組織化
がうまくいっていれば、新しい機能を追加したり古い機能をデバッグしたりす
るのに、システムの局所的部品に⼿を加えるだけでいいようになっているはず
です。
すると、⼤きなプログラムを組織化する⽅法は、モデル化の対象となるシ

ステムを私たちがどう知覚するかというところにかなり左右されることになり
ます。この章では、システムの構造に対する⼆つのかなり異なった “世界観”か
ら浮かび上がってくる、⼆つの顕著な組織化戦略について調べていきます。⼀
つ⽬の組織化戦略はオブジェクト (object)に集中するもので、はっきり分かれ
たオブジェクトの集合として⼤きなシステムを捉え、それらのオブジェクトの
ふるまいは時間とともに変化するとします。もうひとつの組織化戦略は、シス
テムを流れる情報のストリーム (stream)に集中するというものです。これは、
電⼦技術者が信号処理システムを⾒るのと同じやり⽅です。
オブジェクトによるアプローチもストリーム処理によるアプローチも、ど

ちらもプログラムにおける⼤きな⾔語的な問題を引き起こします。オブジェク
トの場合、どうやって計算オブジェクトを変化させながら同⼀性を維持させる
かという問題に取り組む必要があります。この問題のために、古い置換モデル
(1.1.5節)はあきらめて、より機械的なものの理論的には扱いにくい、計算の環
境モデル (environment model)を採⽤することになります。オブジェクト、変
化、同⼀性を扱うことの難しさは、計算モデルの中で時間を扱うことの必要性
から⽣まれる根本的な問題です。プログラムを並⾏実⾏できるようにしようと
すると、これらの問題はますます⼤きくなります。ストリームというアプロー
チは、モデルの中でシミュレートしている時間と、評価の際にコンピュータの
中で起こるイベントの順番とを切り離す場合に最も役に⽴ちます。このことは、
遅延評価 (delayed evaluation)として知られているテクニックによって達成で
きます。

3.1 代⼊と局所状態
私たちは普通、独⽴した物 (オブジェクト)があちこちにあるような場所とし
てこの世界を捉えています。それらのオブジェクトは、時間とともに変わる状
態を持っています。オブジェクトのふるまいがその過去に影響されるとき、そ
のオブジェクトは “状態を持つ” と⾔います。例えば銀⾏⼝座は、“100ドル引
き出せますか?”という質問の答えが預⾦と引き出しの取引履歴によって変わ
るという意味で、状態を持つといえます。オブジェクトの状態は、ひとつ以上
の状態変数 (state variable)によって特徴づけることができます。状態変数は、
オブジェクトの現在のふるまいを決めるのに⼗分な履歴情報を保持します。単
純な銀⾏システムでは、⼝座の取引履歴全体を記憶するのではなく、現在の残
⾼によって特徴づけることができるでしょう。
多くのオブジェクトからなるシステムでは、オブジェクト同⼠が完全に独

⽴しているということはほとんどありません。それぞれのオブジェクトは相互
作⽤によってほかのオブジェクトの状態に影響を与え、その相互作⽤がオブジ
ェクトの状態変数同⼠を結びつけます。実際に、システムがばらばらのオブジ
ェクトによって構成されているという⾒⽅は、システムの状態変数がいくつか
の密結合のサブシステムにグループ化できて、サブシステム同⼠の結合は疎で
あるという場合に最も役に⽴つものです。
このシステムの⾒⽅は、システムの計算モデルを組織化する強⼒な枠組み

となりえます。そのようなモデルをモジュール式にするには、システムの実際
のオブジェクトをモデル化する計算オブジェクトに分解する必要があります。
それぞれの計算オブジェクトは、実際のオブジェクトの状態を表す局所状態変
数 (local state variable)を持たなければなりません。モデル化対象のシステム
に存在するオブジェクトの状態は時間とともに変化するので、それらに対応す
る計算オブジェクトの状態変数も同じく変化しなければなりません。もしシス
テム内での時間の流れをコンピュータ内での経過時間としてモデル化すること
にするなら、プログラムの実⾏に伴ってふるまいの変わる計算オブジェクトを
構築する⽅法が必要です。具体的には、もし状態変数をプログラミング⾔語の
通常の記号名によってモデル化したいならば、プログラミング⾔語は代⼊演算
⼦ (assignment operator)を提供し、名前に関連づけられた値を変化させるこ
とができなければなりません。

3.1.1 局所状態変数
時間とともに変化する状態を持った計算オブジェクトを持つということがど
ういうことなのかを⽰すために、銀⾏⼝座からお⾦を引き出すという状況をモ
デル化することにします。そのために、引き出す amount(⾦額)を引数に取る
withdraw(引き出し)という⼿続きを使います。引き出しを⾏うのに⼗分なお⾦
が⼝座にあれば、withdrawは引き出し後に残る残⾼を返します。そうでなけ
れば、withdrawは Insufficient funds(残⾼不⾜) というメッセージを返します。
例えば、⼝座に 100ドルある状態から始めた場合、withdrawを使うことで、次
のような⼀連の応答を受け取ります。
(withdraw 25)
75
(withdraw 25)
50
(withdraw 60)
"Insufficient funds"
(withdraw 15)
35

式 (withdraw 25)は⼆回評価されていますが、それぞれ違う値を返していると
いうところに注意してください。これは、⼿続きのふるまいとして、これまで
になかった種類のものです。これまでは、⼿続きはすべて数学関数を計算する
仕様として捉えることが可能でした。⼿続きの呼び出しを⾏うと、与えられた
引数を関数に適⽤したときの値の計算が⾏われ、同じ⼿続きを同じ引数で⼆回
呼び出すと、いつでも同じ値が返ってくることになっていました。1

withdraw を実装するために、⼝座の残⾼を⽰す変数 balance を使い、
balanceにアクセスする⼿続きとして withdrawを定義します。withdraw⼿続
きは、balanceが少なくとも要求された amount以上の⼤きさがあるかをチェ
ックします。もしあるなら、withdrawは balanceを amount分だけ減らし、新
しい balanceの値を返します。なければ、withdrawは Insufficient funds(残⾼
不⾜)というメッセージを返します。以下は、balanceと withdrawの定義です。

1実際には、これはあまり正しくありません。例外のひとつとして、1.2.6節の乱数⽣
成器があります。また別の例外としては、2.4.3節で導⼊した演算-型テーブルに関連す
るものがあります。この場合、同じ引数で getを⼆回呼び出した場合の値は、その間の
putの呼び出しによって変わります。⼀⽅、代⼊を導⼊するまで、そのような⼿続きを⾃
分で作る⽅法はないということです。

(define balance 100)
(define (withdraw amount)

(if (>= balance amount)
(begin (set! balance (- balance amount))

balance)
"Insufficient funds"))

balanceの減算は次の式によって⾏います。
(set! balance (- balance amount))

ここでは set!という特殊形式を使っています。構⽂は次のようになります。
(set! ⟨name⟩ ⟨new-value⟩)

ここで、⟨name⟩は記号で、⟨new-value⟩は任意の式です。set!は ⟨name⟩の値を
⟨new-value⟩を評価して得られる結果に変更します。ここでの例では、balance
を変更して、balanceの以前の値から amountを引いた結果を新しい値にしま
す。2

withdrawでは、ifのテストが真の場合に⼆つの式が評価されるようにする
ために、beginという特殊形式も使っています。⼆つの式では、まず balance
を減らし、それから balanceの値を返しています。⼀般的に、以下の式を評価
すると、
(begin ⟨exp1⟩ ⟨exp2⟩ . . . ⟨expk⟩)

⟨exp1 ⟩から ⟨expk ⟩までの式が順番に評価され、最後の式 ⟨expk ⟩の値が begin
形式全体の値として返されます。3

withdrawは意図した通りに動くのですが、変数 balanceが問題になりま
す。上で記述したように、balance はグローバル環境で定義された名前なの
で、どんな⼿続きからでもアクセスして値を調べたり変更したりできてしまい

2set!式の値は実装依存です。set!は作⽤のためだけに使うもので、値のために使う
ものではありません。
set!という名前は、Schemeで使われる命名規則に従っています。変数の値を変更す

る演算 (または、3.3節で学ぶデータ構造を変える演算)は、感嘆符で終わる名前をつけ
ることになっています。これは、述語を指すのに疑問符で終わる名前を使うのと似てい
ます。

3Schemeでは⼿続きの本体として式の連続が使えるので、すでに beginを暗黙的にプ
ログラムの中で使っていることになります。また、cond式の各節の ⟨consequent⟩部分
も、単⼀の式ではなく、式の連続にすることができます。

ます。何とかして balanceを withdrawの内部に置いて、withdrawが balance
に直接アクセスできる唯⼀の⼿続きになるようにして、ほかの⼿続きは全部、
balanceには (withdrawの呼び出しを通して)間接的にしかアクセスできない
ようにできればずっとよくなるでしょう。そうできれば、withdrawが⼝座の状
態を覚えておくために使っている局所状態変数が balanceであるという概念を
より正確にモデル化できることになります。

balanceを withdrawの内部に置くようにするには、定義を以下のように書
き直します。
(define new-withdraw

(let ((balance 100))
(lambda (amount)

(if (>= balance amount)
(begin (set! balance (- balance amount))

balance)
"Insufficient funds"))))

ここでは、初期値 100に束縛された局所変数 balanceを持つ環境を letによっ
て作っています。この局所環境の中で、amountを引数に取り以前の withdraw⼿
続きと同じようなふるまいをする⼿続きを lambdaを使って作っています。let
式を評価した結果として返されるこの⼿続きは new-withdrawで、withdrawと
正確に同じふるまいをしますが、変数 balanceはほかのどの⼿続きからもアク
セスできません。4

set!を局所変数と組み合わせるというのは、今後局所状態を持つ計算オブ
ジェクトを構築するのに使うことになる⼀般的なプログラミングテクニックで
す。残念ながら、このテクニックを使うと深刻な問題が出てきます。最初に⼿
続きを導⼊したとき、⼿続きの適⽤とはどういうことかを解釈するために評価
の置換モデル (1.1.5節)というものも導⼊しました。そのとき、⼿続きの適⽤
とは、仮引数を値で置き換えたうえで⼿続きの本体を評価することとして解釈
するべきだと述べました。問題は、⾔語に代⼊を導⼊したとたん、置換という
のは⼿続き適⽤の適切なモデルではなくなるということです (どうしてそうな

4プログラミング⾔語の業界⽤語では、変数 balance は⼿続き new-withdraw の中
に カプセル化 (encapsulated)されていると⾔います。カプセル化は、隠蔽原理 (hiding
principle)として知られる、⼀般的なシステム設計の原則を反映したものです。隠蔽原理
とは、システムの各部品をほかの部品から守ることによって、システムをよりモジュー
ル的で頑健なものにできるという考え⽅です。これはつまり、“知る必要”のあるシステ
ムの部品にだけ、情報へのアクセス権を与えるということです。

238



るのかは3.1.3節で⾒ていきます)。結果として、厳密に⾔うと、new-withdraw
がなぜ上で述べたようにふるまうのかということについて現時点で理解する⽅
法がないということになります。new-withdrawのような⼿続きをちゃんと理
解するには、⼿続き適⽤についての新しいモデルを開発する必要があります。
3.2節ではそのようなモデルを紹介し、さらに set!と局所変数についての説明
も加えます。しかし、とりあえずは new-withdrawによって打ち⽴てられたテ
ーマについて、いくつかの別のパターンを詳しく⾒ていきましょう。
次の⼿続き make-withdraw は “引き出し処理器” を作成します。make-

withdrawの仮引数 balanceは、⼝座の初期残⾼を指定します。5

(define (make-withdraw balance)
(lambda (amount)

(if (>= balance amount)
(begin (set! balance (- balance amount))

balance)
"Insufficient funds")))

make-withdrawを使うと、以下のように W1と W2という⼆つのオブジェクトを
作ることができます。
(define W1 (make-withdraw 100))
(define W2 (make-withdraw 100))

(W1 50)
50
(W2 70)
30
(W2 40)
"Insufficient funds"
(W1 40)
10

W1 と W2 は完全に独⽴したオブジェクトで、それぞれ⾃分の局所状態変数
balanceを持っているということがわかるでしょうか。⽚⽅から引き出しても、
もう⼀⽅には影響しません。

5上の new-withdrawとは違い、balanceを局所変数にするために letを使う必要はあ
りません。仮引数は元々局所化されているからです。このことは、3.2節で評価の環境モ
デルについて検討すると、よりはっきりします (練習問題 3.10も参照)。

239



引き出しだけでなく預け⼊れも扱うオブジェクトを作ることもできます。
これによって、単純な銀⾏⼝座を表現できます。以下は、指定した初期残⾼を
持つ “銀⾏⼝座オブジェクト”を返す⼿続きです。
(define (make-account balance)

(define (withdraw amount)
(if (>= balance amount)

(begin (set! balance (- balance amount))
balance)

"Insufficient funds"))
(define (deposit amount)

(set! balance (+ balance amount))
balance)

(define (dispatch m)
(cond ((eq? m 'withdraw) withdraw)

((eq? m 'deposit) deposit)
(else (error "Unknown request: MAKE-ACCOUNT"

m))))
dispatch)

make-account を呼び出すたびに、局所状態変数 balance を持つ環境が構築
されます。この環境の中で、make-accountは balanceにアクセスする⼿続き
deposit, withdrawと、また “メッセージ”を⼊⼒として取り⼆つの局所⼿続き
のうちひとつを返す dispatchという追加の⼿続きを定義します。銀⾏⼝座オ
ブジェクトを表現する値としては、dispatch⼿続きそのものが返されます。こ
れはまさに、2.4.3節で学んだ メッセージパッシング (message-passing)のプロ
グラミングスタイルですが、ここでは局所変数を変更する能⼒と合わせて使っ
ています。

make-accountは次のように使えます。
(define acc (make-account 100))
((acc 'withdraw) 50)
50
((acc 'withdraw) 60)
"Insufficient funds"
((acc 'deposit) 40)
90
((acc 'withdraw) 60)

30

accを呼び出すたびに、局所的に定義された depositまたは withdraw⼿続きの
どちらかが返され、それが指定された amountに適⽤されます。make-withdraw
の場合と同じように、make-accountをもう⼀度呼び出すと、
(define acc2 (make-account 100))

完全に別の⼝座オブジェクトを作り、⾃分の局所的な balanceを保持します。
練習問題 3.1: アキュムレータ (accumulator)は、ひとつの数値引
数を伴って繰り返し呼ばれる⼿続きで、引数を合計に集積してい
くというものである。呼び出されるたびに現在までに集積された
合計を返す。それぞれ独⽴した合計を持つアキュムレータを返す
⼿続き make-accumulatorを書け。make-accumulatorへの⼊⼒は、
合計の初期値を指定する。例えば、以下のようになる。
(define A (make-accumulator 5))
(A 10)
15
(A 10)
25

練習問題 3.2: ソフトウェアをテストするアプリケーションでは、
計算の過程で、ある⼿続きが何回呼ばれたかを数えられると便利
だ。⼀引数の⼿続き f を⼊⼒として取る⼿続き make-monitored
を書け。make-monitored が返すのは第三の⼿続き (仮に mf とす
る)で、この⼿続きは内部カウンタを保持することによって呼び
出された回数を覚えておく。mfに対する⼊⼒が how-many-calls?
という特別な記号である場合、mfはカウンタの値を返す。⼊⼒が
reset-countという特別な記号である場合、mfはカウンタをゼロ
にリセットする。それ以外の⼊⼒に対しては、mfはその⼊⼒によ
って fを呼び出した結果を返し、カウンタを 1増やす。例えば、監
視をつけたバージョンの sqrt⼿続きを作ることもできる。
(define s (make-monitored sqrt))
(s 100)
10
(s 'how-many-calls?)
1

練習問題 3.3: make-account⼿続きを修正し、パスワードで守ら
れた⼝座を作成するようにせよ。具体的には、次のように make-
accountが追加の引数としてひとつの記号を取るようにする。
(define acc (make-account 100 'secret-password))

結果としてできる⼝座オブジェクトは、アカウント作成時のパス
ワードを伴っている場合にだけ要求を処理し、その他の場合には
エラーメッセージを出す。
((acc 'secret-password 'withdraw) 40)
60
((acc 'some-other-password 'deposit) 50)
"Incorrect password"

練習問題 3.4: 練習問題 3.3の make-account⼿続きを変更して別
の局所状態変数を追加し、間違ったパスワードで⼝座に 7回連続
アクセスされると call-the-cops(警察を呼ぶ)という⼿続きを呼
ぶようにせよ。

3.1.2 代⼊を導⼊する利点
今後⾒ていくことになりますが、プログラミング⾔語に代⼊を導⼊したことに
よって、我々は難しい概念的問題の藪の中に迷い込むことになります。それで
もなお、局所的な状態を持つオブジェクトの集合としてのシステムの⾒⽅は、
モジュール化された設計を維持していくうえで強⼒なテクニックになります。
単純な例として、rand⼿続きの設計について考えてみましょう。rand⼿続き
は、呼ばれるたびにランダムに選ばれた整数を返します。

“ランダムに選ぶ”というのがどういうことなのかはまったくはっきりして
いません。私たちが望んでいるのは、おそらく、randを連続して呼び出すと⼀
様分布という統計的性質を持つ数列を作るということです。ここでは、適切な
数列を⽣成する⽅法については議論しません。その代わりに、ある数値 x1 か
ら始めると、望ましい統計的性質を持った数列 x1, x2, x3 を以下のように⽣成
するという性質を持つ⼿続き rand-updateがもうあるとします。6

6rand-updateを実装する⼀般的な⽅法としては、a, b, mを適切に選ばれた整数とし
て、xを ax+ b modulo mに更新するというものがあります。Knuth 1981の 3章は、乱
数列を⽣成し、それらの統計的性質を確⽴するテクニックについて広汎に考察していま

x2 = (rand-update x1)
x3 = (rand-update x2)

randは、局所状態変数 xを持ち、それをある固定値 random-initで初期化
する⼿続きとして実装できます。randを呼び出すたびに、現在の xの値に対す
る rand-updateを計算し、それを乱数として返します。また、この値を xの新
しい値として保存します。
(define rand (let ((x random-init))

(lambda ()
(set! x (rand-update x))
x)))

もちろん、単純に rand-updateを直接呼べば、代⼊を使わないで同じ乱数列
を⽣成することもできます。しかし、そうするとプログラムの中で乱数を使
うどの箇所でも、現在の xの値を明⽰的に覚えておいて、rand-updateの引数
として渡さないといけないということになります。これがどれだけ⾯倒なこ
とになるかを実感するために、モンテカルロシミュレーション (Monte Carlo
simulation)と呼ばれるテクニックを実装するために乱数を使う場合について
考えてみましょう。
モンテカルロ法は、⼤きい集合からランダムにサンプルの試⾏を選び、そ

れらの試⾏結果を集計することによって推測した確率をもとに推定を⾏うとい
うものです。例えば、ランダムに選んだ⼆つの整数が共通因⼦を持たない、つ
まり最⼤公約数が 1であるという確率が 6/π2 であるという事実を使って、π
を近似できます。7 π の近似値を求めるためには数多くの試⾏を⾏います。そ
れぞれの試⾏では⼆つの整数をランダムに選び、それらの gcdが 1であるかを
テストします。テストをパスした回数の割合によって 6/π2 の近似値が得られ
るので、この値から πの近似値が得られます。
す。rand-update⼿続きは数学関数を計算しているということに注意してください。同
じ⼊⼒を 2回与えられると、この⼿続きは同じ出⼒を返します。そのため、それぞれの
数値が前の数値と関係していないというのが “ランダム”だとすると、rand-updateに
よって⽣成される数列は、まったく “ランダム”とはいえません。“真のランダム性”と、
はっきりと規定された計算によって⽣成されるものの適切な統計的性質を持つ、いわゆ
る擬似乱数 (pseudo-random)との関係は、数学と哲学の難しい課題が絡む複雑な問題で
す。Kolmogorov, Solomonoff, Chaitinはこれらの問題の解明に⼤きく寄与しました。こ
れに関する議論は Chaitin 1975にあります。

7この定理は E. Cesàroによるものです。議論と証明については、Knuth 1981の 4.5.2
節参照。

プログラムの中⼼となるのは⼿続き monte-carloです。この⼿続きは試⾏
回数と試⾏を引数に取ります。試⾏は、実⾏されるたびに真または偽を返す、
引数なしの⼿続きとして表現されます。monte-carloは、指定された回数だけ
試⾏を実⾏し、試⾏が真となった割合を表す数値を返します。
(define (estimate-pi trials)

(sqrt (/ 6 (monte-carlo trials cesaro-test))))
(define (cesaro-test)

(= (gcd (rand) (rand)) 1))

(define (monte-carlo trials experiment)
(define (iter trials-remaining trials-passed)

(cond ((= trials-remaining 0)
(/ trials-passed trials))

((experiment)
(iter (- trials-remaining 1)

(+ trials-passed 1)))
(else
(iter (- trials-remaining 1)

trials-passed))))
(iter trials 0))

今度は、randの代わりに rand-updateを使って同じ計算をしてみましょう。局
所状態をモデル化するために代⼊を使わないとすると、このやり⽅をするしか
ありません。
(define (estimate-pi trials)

(sqrt (/ 6 (random-gcd-test trials random-init))))
(define (random-gcd-test trials initial-x)

(define (iter trials-remaining trials-passed x)
(let ((x1 (rand-update x)))

(let ((x2 (rand-update x1)))
(cond ((= trials-remaining 0)

(/ trials-passed trials))
((= (gcd x1 x2) 1)
(iter (- trials-remaining 1)

(+ trials-passed 1)
x2))

(else
(iter (- trials-remaining 1)

trials-passed
x2))))))

(iter trials 0 initial-x))

プログラムはまだ単純ではありますが、残念ながらモジュール性が破られて
しまっているところがいくつかあります。プログラムの最初のバージョンで
は、randを使って、任意の experiment⼿続きを引数として取る汎⽤の monte-
carlo⼿続きとしてモンテカルロ法を直接表現できました。プログラムの⼆つ
⽬のバージョンでは、乱数⽣成器の局所状態がないので、random-gcd-testは
乱数 x1と x2を明⽰的に操作し、rand-updateに対する新たな⼊⼒として x2
を繰り返しのループの間ずっと再利⽤する必要があります。このように乱数を
明⽰的に扱うと、テスト結果を集積していくという構造と、この特定の試⾏で
は乱数を⼆つ使うという事実とが結びついてしまいます。しかし、ほかのモン
テカルロ法の試⾏では、使う乱数の数が⼀つだったり三つだったりするかもし
れません。トップレベルの⼿続きである estimate-piまで、乱数の初期値を提
供することについて⾯倒を⾒なければなりません。乱数⽣成器の内部がプログ
ラムのほかの部分に漏れてしまっているため、モンテカルロ法という考え⽅を
分離して、ほかのタスクにも適⽤できるようにするということが難しくなって
います。プログラムの最初のバージョンでは、代⼊が乱数⽣成器の状態を rand
⼿続きの中にカプセル化しているので、乱数⽣成の細かいところはプログラム
のほかの部分から独⽴していることができます。
モンテカルロ法の例から、次の⼀般的な現象がわかります。複雑なプロセ

スのある部品という⽴場から⾒ると、ほかの部品は時間とともに変化している
ように⾒えます。それらの部品は、時とともに変化する隠れた局所状態を持っ
ています。このような分解⽅法を反映した構造を持つコンピュータプログラム
を書こうと思うと、ふるまいが時間とともに変わる計算オブジェクト (銀⾏⼝
座や乱数⽣成器など)を作ることになります。状態は局所状態変数によってモ
デル化し、状態の変化はそれらの変数への代⼊としてモデル化します。
こうなると、次のように結論づけてしまいたくなるところです。代⼊と、状

態を局所変数に隠すというテクニックを導⼊したことで、追加の引数を渡すこ
とによってすべての状態を明⽰的に操作するよりも、よりモジュール性の⾼い
スタイルでシステムを構造化できるようになった、と。残念ながら、これから
⾒ていくように、話はそんなに簡単ではありません。

練習問題 3.5: モンテカルロ積分 (Monte Carlo integration)は、モ

245



ンテカルロシミュレーションによって定積分を推定する⼿法であ
る。領域内の点 (x, y)に対しては真となり、領域外の点に対しては
偽となるような述語 P (x, y)によって記述される領域の⾯積を計
算する場合について考える。例えば、中⼼ (5, 7)で半径が 3であ
る円に含まれる領域は、(x − 5)2 + (y − 7)2 ≤ 32 となるかどうか
をテストする述語によって記述される。そのような述語によって
記述された領域の⾯積を推定するには、まずその領域を包含する
⻑⽅形を選ぶ。例えば、向かい合う頂点が (2, 4)と (8, 10)にある
⻑⽅形は、上で述べた円を含む。求める積分は、⻑⽅形の中でそ
の領域の中に⼊る部分の⾯積である。その積分は、ランダムに⻑
⽅形の中の点 (x, y)を選んでいき、それぞれの点について P (x, y)

をテストし、その点が元の領域の中にあるかどうかを決定するこ
とによって推定できる。もしこの試⾏を多くの点について⾏えば、
領域の中に⼊る点の割合から、その領域の中にある⻑⽅形の割合
を推定することができる。よって、その割合に⻑⽅形全体の⾯積
をかけると、積分の推定値が得られることになる。
モンテカルロ積分を、⼿続き estimate-integralとして実装せよ。
引数として、述語 P、⻑⽅形の上下界 x1, x2, y1, y2、推定値を⽣
成するための試⾏回数を取る。この⼿続きは、上で π を推定する
ために使ったものと同じ monte-carlo⼿続きを使わなければなら
ない。この estimate-integralを使って、単位円の⾯積を測るこ
とによって πの推定値を求めよ。
この⼿続きを書くのには、与えられた範囲からランダムに選んだ
数値を返す⼿続きがあれば便利だ。以下の random-in-range⼿続
きは、1.2.6節で使った、⼊⼒よりも⼩さい⾮負数を返す random⼿
続きによってそれを実装している。8

(define (random-in-range low high)
(let ((range (- high low)))

(+ low (random range))))

練習問題 3.6: 乱数⽣成器をリセットして、与えられた値から始
まる数列を作れるようにできると便利だ。記号 generate または

8mit Schemeはそのような⼿続きを提供しています。randomに (1.2.6節のように)整
数が渡されれば整数を返しますが、(この課題のように) ⼩数を渡されると⼩数を返しま
す。

resetのどちらかを引数として取り、次のようなふるまいをする新
しい rand⼿続きを設計せよ。(rand 'generate)は新しい乱数を
⽣成する。((rand 'reset) ⟨new-value ⟩)は内部の状態変数を指定
された ⟨new-value⟩にリセットする。つまり、状態をリセットする
ことによって、再現可能な数列が⽣成できる。これは乱数を使うプ
ログラムをテストしたりデバッグしたりするのにとても役に⽴つ。

3.1.3 代⼊を導⼊することのコスト
ここまでで⾒てきたように、set!演算があると局所状態を持つオブジェクト
をモデル化できるようになります。しかし、この利点は代償を伴うものです。
このことによって、私たちのプログラミング⾔語は1.1.5節で紹介した⼿続き
適⽤の置換モデルによって解釈することがもうできなくなっています。それだ
けでなく、プログラミング⾔語のオブジェクトと代⼊を扱う適切な枠組みとし
て使えるような、“いい感じの”数学的性質を持った単純なモデルというものは
存在しません。
代⼊を使わない限り、同じ引数による同じ⼿続きの呼び出しは⼆回評価して

も同じ結果になるので、⼿続きは数学関数の計算と⾒なすことができます。この
本の⼀章と⼆章を通してしてきたような、まったく代⼊を使わないプログラミ
ングというのは、そのために関数型プログラミング (functional programming)
と呼ばれています。
代⼊によってどれだけ話が難しくなるかということを、3.1.1 節の make-

withdraw ⼿続きを単純化したバージョンを使って考えてみましょう。このバ
ージョンでは、残額が⼗分かどうかのチェックをしないようにしています。
(define (make-simplified-withdraw balance)

(lambda (amount)
(set! balance (- balance amount))
balance))

(define W (make-simplified-withdraw 25))
(W 20)
5
(W 10)
-5

この⼿続きと、set!を使わない次の make-decrementer⼿続きとを⽐べてみて
ください。

(define (make-decrementer balance)
(lambda (amount)

(- balance amount)))

make-decrementerは、指定された残⾼ balanceから⼊⼒値を引く⼿続きを返
します。しかし、続けて呼び出しても、make-simplified-withdrawのような
集積効果はありません。
(define D (make-decrementer 25))
(D 20)
5
(D 10)
15

make-decrementerの動作の説明には置換モデルが使えます。例えば、以下の
式の評価を解析してみましょう。
((make-decrementer 25) 20)

まず、make-decrementer本体の balanceを 25によって置き換えて、組み合わ
せの演算⼦を単純にします。式は以下のようになります。
((lambda (amount) (- 25 amount)) 20)

lambda式本体にある amountを 20によって置き換えて、演算⼦を適⽤します。
(- 25 20)

最終的な答えは 5になります。
しかし、make-simplified-withdrawに対して同じような置換による解析

をしようとするとどうなるでしょうか。
((make-simplified-withdraw 25) 20)

まず、make-simplified-withdrawの本体にある balanceを 25に置き換えて
演算⼦を単純にします。すると、式は次のようになります。9

((lambda (amount) (set! balance (- 25 amount)) 25) 20)

今度は、lambda式本体の amountを 20に置き換えて演算⼦を適⽤します。
9set!式に出てくる balanceは置き換えません。set!内の ⟨name⟩は評価されてい

ないからです。これを置き換えるとすると、(set! 25 (- 25 amount))というおかしな
結果になってしまいます。

(set! balance (- 25 20)) 25

置換モデルにあくまでこだわるのであれば、この⼿続き適⽤の意味は、まず
balanceを 5に設定して、それから式の値として 25を返すというものだ、と
⾔わなければならないことになります。これは間違った答えです。正しい答
えを得るには、何とかして⼀つ⽬の balance(set! の効果の前) と⼆つ⽬の
balance(set!の効果の後)とを区別する必要があります。しかし、これは置換
モデルではできません。
ここで問題になるのは、置換というものが究極的に、私たちの⾔語の記号

というものが本質的に値に対する名前であるという概念に基づいているという
ことです。しかし、set!を導⼊し、変数の値は変えられるものだという考え⽅
を導⼊したとたん、変数はもはや単純に名前であるとはいえなくなります。変
数は、値が格納される場所を何らかの⽅法で参照するもので、その場所に格納
された値は変えることができるということになります。3.2節では、私たちの
計算モデルの中で、環境がこの “場所”という役割を勤めるということを⾒て
いきます。

同⼀性と変化
ここで浮上してきた問題は、ある特定の計算モデルが破綻するというだけでは
なく、より奥の深いものです。私たちの計算モデルに変更を導⼊したとたん、
これまで単純だった多くの概念に問題が出てきます。例えば、⼆つのものが
“同じ”であるという概念について考えてみます。
同じ引数で make-decrementerを⼆回呼び、⼆つの⼿続きを作成するとし

ます。
(define D1 (make-decrementer 25))
(define D2 (make-decrementer 25))

D1と D2は同じでしょうか。無難な答えは「同じ」です。D1と D2は同じ計算上
のふるまい—どちらも⼊⼒から 25を引くという⼿続きだということ—を持っ
ているからです。実際、どんな計算であっても、結果を変えることなく D2の
代わりに D1を使うことができます。
これと、make-simplified-withdrawを⼆回呼ぶ場合とを対⽐してみまし

ょう。
(define W1 (make-simplified-withdraw 25))
(define W2 (make-simplified-withdraw 25))

W1と W2は同じでしょうか。これは確実に違います。W1と W2の呼び出しは、次
の対話でわかるように、異なる効果を持っています。
(W1 20)
5
(W1 20)
-15
(W2 20)
5

W1と W2は、(make-simplified-withdraw 25)という同じ式を評価して作られ
たという意味では “同じ”ものですが、どんな式であっても、式の評価結果を変
えずに W2の代わりに W1が使えるということはありません。
式の値を変化させることなく、式の中で “等しいものは等しいものによっ

て置き換えることができる” という概念をサポートしている⾔語は、参照透明
(referentially transparent) であると⾔われます。私たちのコンピュータ⾔語に
set!を含めたときに、参照透明性は破られました。そのため、どういう場合に
等価な式を置き換えることによって式を単純にできるかということを決めるの
は厄介な問題になります。結果として、代⼊を使うプログラムについて推論を
⾏うことは劇的に難しくなります。
参照透明性をあきらめると、計算オブジェクトが “同じ”であるというのは

どういう意味かという概念は、形式的に捉えることが難しくなります。実際の
ところ、私たちのプログラムのモデル化対象である現実世界において “同じ”
というのがどういう意味かということ⾃⾝、はっきりしているとはとても⾔え
ません。⼀般的に、⾒たところ同⼀に⾒えるオブジェクトが実際に “同じもの”
であるかどうかは、⽚⽅のオブジェクトを変更してみて、もう⼀⽅のオブジェ
クトも同じように変更されるかどうかを観測することによって決めるしかあり
ません。しかし、あるオブジェクトが “変化”したかどうかを調べるには、“同
じ”オブジェクトを⼆回観測して、そのオブジェクトの属性が最初の観測と⼆
回⽬の観測で違うかどうかを調べる以外に⽅法がありません。このため、“変
化”を測定することは、“同じ”ということについての a priori(先験的)な概念
なしにはできません。そして、同じかどうかを測定することは、変化の効果を
観測することなしにはできません。
プログラミングの中でこの問題が起こってくる例として、Peterと Paulが

100ドル⼊っている⼝座を持っているという状況について考えます。これを次
のようにモデル化するのと、
(define peter-acc (make-account 100))

(define paul-acc (make-account 100))

次のようにモデル化するのとでは、⼤きな違いがあります。
(define peter-acc (make-account 100))
(define paul-acc peter-acc)

⼀つ⽬の状況では、⼆つの銀⾏⼝座は違うものです。Peterによる取引は Paul
の⼝座には影響せず、その逆も同じです。しかし、⼆つ⽬の状況では、paul-
accは peter-accと 同じものになるよう定義しています。実質的に、この場合
は Peterと Paulが共同⼝座を持っていることになり、Peterが peter-accから
引き出しをすると、Paulは paul-accの残額が減っていることを観測すること
になります。これら⼆つの、似ているものの異なっている状況は、計算モデル
を構築するうえで混乱を招きます。共同⼝座があると、具体的には、ひとつの
オブジェクト (銀⾏⼝座)で⼆つの異なる名前 (peter-accと paul-acc)を持つ
ものがあるということが特に混乱を招くものになります。仮に、プログラムの
中で paul-accが変更される可能性がある場所をすべて検索しようとしている
とすると、peter-accを変更するものも忘れずに探さなければなりません。10

“同じ”、“変化”について上で述べたことに関して、もし Peterと Paulが銀
⾏の残⾼を調べることしかできず、残⾼を変えられるような操作ができないと
すると、⼆つの⼝座が異なるかどうかという問題は意味がなくなるということ
に注意してください。⼀般的に、もしデータオブジェクトを決して変更しない
のであれば、複合データオブジェクトはその部品を合わせたものだと⾒なすこ
とができます。例えば、有理数は分⺟と分⼦によって決定されます。しかしこ
の⾒⽅は、変化がある場合には有効でなくなります。その場合、複合データは、
複合データを構成する部品を合わせたものとは少し違う “アイデンティティ”
を持つことになります。銀⾏⼝座は、引き出しを⾏って残⾼を変更しても、依
然として “同じ”銀⾏⼝座です。その反対に、同じ状態情報を持つ⼆つの異なる
銀⾏⼝座を作ることもできます。この複雑性の原因となっているのは、私たち

10ひとつの計算オブジェクトが⼆つ以上の名前によってアクセスされるという現象は、
エイリアス (aliasing)と呼ばれます。共同⼝座の状況は、エイリアスのとても単純な例
を⽰しています。3.3節では、“別の”複合データ構造が部品を共有するといった、より
複雑な例について学びます。もし、⼆つの “異なる”オブジェクトが実際はひとつのオ
ブジェクトに異なるエイリアスがついたものであることによって、あるオブジェクトを
変更すると “副作⽤”として “異なる”オブジェクトを変更してしまうことがあるという
ことを忘れていると、プログラムにバグが起こったりします。このような、いわゆる副
作⽤バグ (side-effect bug)というものは、特定が難しく解析も難しいため、プログラミ
ング⾔語は副作⽤やエイリアスを許可しないよう設計するべきだと提案する⼈もいます。
(Lampson et al. 1981; Morris et al. 1980)

のプログラミング⾔語ではなく、私たちが銀⾏⼝座をオブジェクトとして捉え
ているということです。例えば、有理数については、分⼦を変更しても “同じ”
有理数であり続けることができるようなアイデンティティを持つ変更可能なオ
ブジェクトと⾒なしたりすることは普通ありません。

命令型プログラミングの落とし⽳
関数型プログラミングとは対照的に、代⼊を多⽤するプログラミングは命令型
プログラミング (imperative programming)と⾔います。計算モデルのいろいろ
な問題を引き起こすだけでなく、命令型スタイルで書いたプログラムは、関数
型プログラムでは起こり得ないバグが起こりやすくなります。例えば、1.2.1
節の反復階乗プログラムのことを思い出してみましょう。
(define (factorial n)

(define (iter product counter)
(if (> counter n)

product
(iter (* counter product) (+ counter 1))))

(iter 1 1))

内部の反復ループで、引数を渡す代わりに、変数 productと counterの値を更
新するのに明⽰的な代⼊を使うことで、より命令型のスタイルにできます。
(define (factorial n)

(let ((product 1)
(counter 1))

(define (iter)
(if (> counter n)

product
(begin (set! product (* counter product))

(set! counter (+ counter 1))
(iter))))

(iter)))

これは、プログラムの⽣成する結果を変えるわけではありません。しかし、こ
こで微妙な罠が⼊り込んでいます。代⼊の順序はどうやって決めたらいいので
しょうか。今回の場合、プログラムはたまたまこのままで正しいものになって
います。しかし、代⼊を逆の順番で書いてしまうと、

(set! counter (+ counter 1))
(set! product (* counter product))

別の、間違った答えになるところです。⼀般的に、代⼊のあるプログラミング
をする際には、代⼊の相対的な順番を注意深く考えて、それぞれの⽂が、中⾝
の変わった各変数の正しいバージョンを使うように気をつけないといけませ
ん。この問題は、関数型プログラミングでは最初から起こりません。11

命令型プログラミングの複雑性は、複数のプロセスが並⾏に実⾏されるア
プリケーションを考えると、ますますひどいことになります。この点について
は3.4節で⾒ていくことにします。ここでは、まずは代⼊を含む式について計
算モデルを与えるという問題に取り組み、またシミュレーションの設計をする
のに局所状態を持つオブジェクトを使うことについて検討します。

練習問題 3.7: 練習問題 3.3で述べた、パスワード機能を追加した
make-accountで作った銀⾏⼝座オブジェクトについて考える。私
たちの銀⾏システムでは、共同⼝座を作る能⼒が必要だとする。こ
れを実現する⼿続き make-jointを定義せよ。make-jointは三つ
の引数を取る。⼀つ⽬はパスワード保護された⼝座である。⼆つ
⽬の引数は⼝座を定義したときのパスワードと⼀致している必要
があり、そうでなければ make-joint演算を進めることはできな
い。三番⽬の引数は新しいパスワードである。make-jointは、元
の⼝座に新しいパスワードでもアクセスできるようにする。例え
ば、peter-accが open-sesameというパスワードを持つ銀⾏⼝座
だとすると、
(define paul-acc

(make-joint peter-acc 'open-sesame 'rosebud))

このようにすることで、paul-accという名前と rosebudというパ
スワードによって peter-acc に対する取引ができるようにする。

11このことを考えると、初歩的なプログラミングを教える際に、⾮常に命令型に偏った
スタイルがよく使われるというのは⽪⾁なことです。これは、1960年代と 1970年代を
通して⼀般的だった、⼿続きを呼ぶプログラムは代⼊を⾏うプログラムに⽐べて本質的
に⾮効率的だ (Steele 1977はそれが誤りだと証明しています)という信念の名残かもしれ
ません。あるいは、初⼼者にとってはステップごとの代⼊のほうが⼿続きの呼び出しよ
りもイメージしやすいという考え⽅によるものかもしれません。理由はどうあれ、この
ことによって初級プログラマに “この変数の設定は、あの変数より前にしたほうがいい
のか、後にしたほうがいいのか”という⼼配をさせることになり、プログラミングを複雑
にして、重要な考え⽅を⾒えにくくさせてしまうということがよくあります。

この新しい機能に対応するには、練習問題 3.3の解答を修正するの
がよいかもしれない。
練習問題 3.8: 1.1.3節で評価モデルを定義したときに、式を評価す
る最初のステップはその部分式を評価することだと述べた。しか
し、部分式を評価する順番 (例えば、左から右、または右から左)
については規定しなかった。代⼊を導⼊すると、⼿続きの引数を
評価する順番によって結果が変わるということが起こりえる。単
純な⼿続き fを定義して、次の式
(+ (f 0) (f 1))

を評価する際に、+の引数の評価順が左から右であれば 0を返し、
右から左であれば 1を返すようにせよ。

3.2 評価の環境モデル
第 1章で複合⼿続きを導⼊した際には、⼿続きを引数に適⽤するということの
意味を定義するため、評価の置換モデル (1.1.5節)を使いました。

• 複合⼿続きを引数に適⽤するには、⼿続きの本体に出てくる仮引数を対
応する引数で置き換えて、それを評価する。

プログラミング⾔語で代⼊を認めることにすると、このような定義はもはや適
切ではなくなります。特に、3.1.3節では、代⼊があると変数はもはや単なる値
に対する名前であるとは⾔えなくなるということを⽰しました。変数は、値が
格納できる “場所”を何らかの⽅法で指していなければならないことになりま
す。新しい評価モデルでは、これらの場所は環境 (environment) と呼ばれる構
造で保存されます。
環境というのはフレーム (frame)が並んだものです。それぞれのフレーム

は、変数の名前と、それに対応する値を結びつける束縛 (binding) のテーブ
ル (空のこともある)です (ひとつのフレームは、任意の変数に対して⾼々ひ
とつの束縛を持っています)。それぞれのフレームは、そのほかに外側の環境
(enclosing environment)へのポインタも持っています。例外として、ここで説
明のためにグローバル (global)と呼ぶものは外側の環境へのポインタを持ちま
せん。ある環境についての変数の値 (value of a variable)とは、環境の中でそ
の変数に対する束縛を含む最初のフレームの中で、その変数の束縛によって
与えられる値ということになります。もしフレームの列の中にその変数に対

C D

z:6
x:7

m:1
y:2

x:3
y:5

図 3.1: 単純な環境構造

する束縛を規定しているものがなければ、その変数はその環境の中で未束縛
(unbound) であると⾔います。
図 3.1は、三つのフレームからなる簡単な環境の構造を⽰したものです。フ

レームにはそれぞれ、I, II, IIIというラベルがついています。この図の中で、A,
B, C, Dは環境へのポインタです。Cと Dは同じ環境を指しています。変数 z
と xがフレーム IIに束縛されていて、yと xがフレーム Iに束縛されています。
環境 Dの中では、xの値は 3です。環境 Bについても、xの値は 3です。これ
は次のように決定されています。まず、列の最初のフレーム (フレーム III)を
調べます。xに対する束縛は⾒つからないので、外側の環境 Dに進み、フレー
ム Iで束縛が⾒つかります。⼀⽅、環境 Aでは xの値は 7です。列の最初のフ
レーム (フレーム II)が xと 7の束縛を含んでいるからです。環境 Aについて、
フレーム IIの xと 7の束縛が、フレーム Iの xと 3の束縛を隠蔽 (shadow)し
ていると⾔います。
環境は評価プロセスにとって不可⽋なものです。式を評価するコンテキス

トを決めるのは環境だからです。プログラミング⾔語の式は、それ⾃⾝意味を
持たないとも⾔えます。式は、式が評価される環境があって初めて意味を持つ
ことになります。(+ 1 1)のような単純な式を解釈するのにも、+が加算の記
号だというコンテキストの中で演算をしていると理解していることが必要にな
ります。そのため、私たちの評価モデルの中では、式の評価について語るとき
には、必ず環境についても語ることにします。また、インタプリタとの対話に
ついて説明するために、グローバル環境というものがあると想定します。グロ
ーバル環境は (外側の環境のない)単⼀のフレームからなり、そのフレームは記

255



号の値として基本⼿続きとの関連づけを含んでいるとします。例えば、+が加
算の記号であるということは、記号 +はグローバル環境の中で基本要素の加算
⼿続きに束縛されているというように捉えます。

3.2.1 評価規則
インタプリタが組み合わせを評価するやり⽅に関する全体的な仕様は、最初
に1.1.3節で導⼊したものと同じものになります。

• 組み合わせを評価するには、
1. 組み合わせの部分式を評価する。12

2. 演算⼦となっている部分式の値を、被演算⼦となっている部分式の値に
適⽤する。

評価の環境モデルでは、複合⼿続きを引数に適⽤するということの意味の規定
の仕⽅が置換モデルと違います。
評価の環境モデルでは、⼿続きは常に、何らかのコードと、ある環境への

ポインタというペアになります。⼿続きを作る⽅法はひとつだけです。その⽅
法とは、λ式を評価することです。それによって⼿続きが作られ、λ式のテキ
ストがその⼿続きのコードとなり、λ式が評価されたときの環境がその⼿続き
の環境になります。例えば、次の⼿続きの定義がグローバル環境の中で評価さ
れる場合について考えてみましょう。
(define (square x)

(* x x))

この⼿続き定義の⽂法は、その基層にある暗黙の λ式に対するシンタックスシ
ュガーにすぎません。次のものを使っても同じことです。
(define square

(lambda (x) (* x x)))

12代⼊があると、ステップ 1に難しいところが出てきます。練習問題 3.8で⽰したよう
に、代⼊があると、組み合わせの部分式がどういう順番で評価されるかによって違う値
を返すような式が書けるようになるからです。そのため、正確には、ステップ 1の評価
順序 (例えば、左から右や右から左)を規定しておく必要があります。しかし、この順序
は常に実装の細部として捉えられるべきものであって、特定の順序に依存したプログラ
ムは決して書いてはいけません。例えば、よくできたコンパイラであれば、部分式の評
価順序を変えることによってプログラムを最適化したりすることもありえるからです。

other variables

square:

global
env

(define (square x)
  (* x x))

parameters: x

body: (* x x)

図 3.2: グローバル環境の中で (define (square x) (* x
x))を評価することによって作られる環境

これは、グローバル環境の中で (lambda (x) (* x x))を評価し、グローバル
環境の中で square とその結果を束縛するものです。
図 3.2は、この define式を評価した結果を⽰したものです。⼿続きオブジ

ェクトはペアであり、そのコード部分は、⼿続きが xという仮引数をひとつ持
つということと、⼿続きの本体が (* x x)であるということを規定しています。
⼿続きの環境部分はグローバル環境へのポインタになります。λ式を評価して
⼿続きを作ったのがグローバル環境の中だったからです。⼿続きオブジェクト
と記号 squareを結びつける束縛は、グローバルなフレームに追加されていま
す。⼀般的に、defineはフレームに束縛を追加することによって定義を作成し
ます。
⼿続きの作られ⽅がわかったので、⼿続きがどのように適⽤されるかとい

うことも説明できるようになります。環境モデルは、以下のことを規定します。
⼿続きを引数に適⽤するには、仮引数を引数の値に束縛するフレームを含む新
しい環境を作ります。このフレームの外側の環境は、⼿続きによって規定され
る環境です。そして、この新しい環境の中で、⼿続きの本体を評価します。
この規則が実際にどのように適⽤されるか⾒てみましょう。図 3.3は、グロ

ーバル環境で (square 5)という式を評価することによって作られる環境構造
を⽰したものです。ここで、squareは図 3.2で⽣成された⼿続きです。⼿続き
を適⽤すると新しい環境が作られます。図の中では E1というラベルをつけて
います。この環境の最初のフレームでは、⼿続きの仮引数である xが、引数の 5

E1

(* x x)
parameters: x

body: (* x x)

(square 5)

global
env

other variables

square:

x:5

図 3.3: グローバル環境で (square 5)を評価することによ
って作られる環境

に束縛されています。このフレームから上に伸びている⽮印のポインタは、こ
のフレームの外側の環境がグローバル環境であるということを⽰しています。
ここでグローバル環境が選ばれているのは、square⼿続きオブジェクトの環
境部分が指しているのがグローバル環境だからです。E1の中で、⼿続きの本体
である (* x x)を評価します。E1の中での xの値は 5であるため、結果は (*
5 5)、つまり 25になります。
⼿続き適⽤の環境モデルを要約すると、次の⼆つの規則になります。
• 引数セットへの⼿続きオブジェクトの適⽤は、フレームを構築し、⼿続
きの仮引数を呼び出しの引数に束縛し、それから新しく構築された環境
という⽂脈の中で⼿続きの本体を評価することによって⾏う。新しいフ
レームは、適⽤する⼿続きオブジェクトの環境部分を外側の環境として
持つ。

• ⼿続きは、与えられた環境について λ式を評価することによって作る。
結果としてできる⼿続きオブジェクトは、λ式のテキストと、⼿続きが
作成された環境へのポインタからなるペアである。

このほかに、defineを使って記号を定義すると、現在の環境フレームの中に束
縛を作り、その記号に指定の値を割り当てるということも規定します。13 最後

13もしその変数に対する束縛が現在のフレームにすでにあれば、その束縛は変更され
ます。こうすると記号が再定義できるので便利です。しかし、それができるということ

に、set!のふるまいを規定します。そもそも、環境モデルを導⼊しなければなら
なくなったのはこの演算のせいでした。ある環境で、(set! ⟨variable⟩ ⟨value⟩)
という式を評価すると、その環境の中でのその変数に対する束縛を探し、その
束縛を変更して新しい値を指すようにします。つまり、その変数に対する束縛
を含む、その環境の中での最初のフレームを探し、そのフレームを修正すると
いうことです。もしその変数がその環境の中で束縛されていなければ、set!は
エラーを発⽣させます。
これらの評価規則は、置換モデルよりはだいぶ複雑ですが、まだそれなり

に単純です。それに、この評価モデルは抽象的ではあるものの、インタプリタ
が式を評価するやり⽅を正しく描写しています。第 4章では、ちゃんと動くイ
ンタプリタを実装するのに、このモデルが⻘写真となってくれるということを
⾒ていきます。次の節では、いくつかの具体例としてのプログラムを解析して
いくことで、このモデルの詳細について解説していきます。

3.2.2 単純な⼿続きの適⽤
1.1.5節で置換モデルを導⼊した際に、以下のような⼿続き定義があるとき、(f
5)という組み合わせを評価した結果がどのようにして 136になるかというこ
とを⽰しました。
(define (square x)

(* x x))
(define (sum-of-squares x y)

(+ (square x) (square y)))
(define (f a)

(sum-of-squares (+ a 1) (* a 2)))

同じ例を環境モデルを使って分析することもできます。図 3.4は、グローバル
環境で f, square, sum-of-squaresの定義を評価することによって作られる三
つの⼿続きオブジェクトを⽰したものです。それぞれの⼿続きオブジェクトは、
何かしらのコードとグローバル環境へのポインタから構成されています。
図 3.5には、式 (f 5)を評価することによって作られる環境構造を⽰してあ

ります。fを呼び出すことによって E1という新しい環境が作られ、その中のフ
は、defineを使って値の変更ができるということになります。つまり、明⽰的に set!を
使っていないのに、代⼊に関する問題が出てくることになります。このため、すでに存
在する記号を再定義しようとするとエラーや警告を出すようにしてほしいという⼈もい
ます。

square:

sum-of-squares:

f:

global
env

parameters: a

body: (sum-of-squares
       (+ a 1)
       (* a 2))

parameters: x

body: (* x x)

parameters: x,y

body: (+ (square x)
        (square y))

図 3.4: グローバルフレーム内の⼿続きオブジェクト

レームで fの仮引数である aが引数の 5に束縛されます。E1の中で、fの本体
を評価します。
(sum-of-squares (+ a 1) (* a 2))

この組み合わせを評価するには、まず部分式から評価します。最初の部分式
sum-of-squaresは、値として⼿続きオブジェクトを持っています (この値の⾒
つけ⽅に注意してください。まず E1の最初のフレームを⾒ますが、そこには
sum-of-squaresに対する束縛がありません。次に、外側の環境、つまりグロー
バル環境に進み、図 3.4に⽰した束縛が⾒つかります)。残り⼆つの部分式の評
価は、基本演算 +と *を適⽤して (+ a 1)と (* a 2)という⼆つの組み合わせ
を評価することによって⾏えます。結果はそれぞれ 6, 10になります。
今度は、⼿続きオブジェクト sum-of-squaresを引数 6, 10に適⽤します。

その結果、新しい環境 E2が作られ、その中で仮引数 x, yがそれらの引数に束
縛されます。E2の中で、(+ (square x) (square y))という組み合わせを評
価します。すると、次は (square x)を評価することになります。squareはグ
ローバルフレームで⾒つかり、xは 6です。もう⼀度新しい環境 E3を⽤意し
て、xを 6に束縛し、E3の中で squareの本体 (* x x)を評価します。また、
sum-of-squaresの適⽤の中では、(square y)という部分式も評価する必要が
あります。ここで、yは 10です。この⼆回⽬の squareの呼び出しでは、また
別の環境 E4が作られ、その中で squareの仮引数 xが 10に束縛されます。そ
して、E4の中で (* x x)を評価しなければなりません。

(* x x)

x:10
E4

(* x x)

x:6
E3

(+ (square x)
   (square y))

x:6
y:10E2

(sum-of-squares
  (+ a 1)
  (* a 2))

a:5
E1

(f 5)

global
env

図 3.5: 図 3.4内の⼿続きを⽤いて (f 5)を評価することで
作られた環境

注意すべき重要な点は、squareを呼び出すたびに、xの束縛を含む新しい
環境が作られるということです。ここから、それぞれのフレームが、どれも x
という名前を持つ異なる局所変数をそれぞれ別々にしておくのに役に⽴ってい
るということがわかります。squareによって作られたそれぞれのフレームが
グローバル環境を指していることに注意してください。これは、squareの⼿続
きオブジェクトが指している環境がグローバル環境だからです。
部分式が評価された後には、その結果が返されます。squareの⼆回の呼び

出しによって⽣成された値は、sum-of-squaresによって加算され、その結果
が fによって返されます。ここで私たちが着⽬しているのは環境構造なので、
これらの返される値がどのように複数の呼び出しの間で渡されていくかという
ところには深⼊りしません。しかし、これも評価プロセスの重要な側⾯のひと
つです。この点については、第 5章で詳しく⾒ていきます。

練習問題 3.9: 1.2.1節では、階乗を計算する⼆つの⼿続きを解析す
るのに置換モデルを使った。再帰版は以下の通りで、
(define (factorial n)

(if (= n 1) 1 (* n (factorial (- n 1)))))

反復版は以下の通りである。

(define (factorial n) (fact-iter 1 1 n))
(define (fact-iter product counter max-count)

(if (> counter max-count)
product
(fact-iter (* counter product)

(+ counter 1)
max-count)))

それぞれの factorial⼿続きを使って (factorial 6)を評価する
ことによって作られる環境構造を⽰せ。14

3.2.3 局所状態の保管場所としてのフレーム
局所状態を持つオブジェクトを表現するのに⼿続きと代⼊を使うやり⽅につい
て考えるのには環境モデルが役に⽴ちます。ひとつの例として、次の⼿続きを
呼び出すことによって作られる 3.1.1節の “引き出し処理器”について考えてみ
ましょう。
(define (make-withdraw balance)

(lambda (amount)
(if (>= balance amount)

(begin (set! balance (- balance amount))
balance)

"Insufficient funds")))

次の式を評価し、
(define W1 (make-withdraw 100))

続けて次の式を評価する場合について説明します。
(W1 50)
50

図 3.6は、グローバル環境で make-withdraw⼿続きを定義した結果を⽰してい
ます。この定義は、グローバル環境へのポインタを持つ⼿続きオブジェクトを

141.2.1節で述べた、インタプリタは fact-iterのような⼿続きを末尾再帰を使うこと
によって⼀定の空間量で実⾏できるという主張は、環境モデルでは曖昧にしておきます。
末尾再帰については、5.4節でインタプリタの制御構造を扱う際に考察します。

parameters: balance

body: (lambda (amount)
        (if (>= balance amount)
            (begin (set! balance (-- balance amount))
                   balance)
            "insufficient funds"))

global
env

make-withdraw:

図 3.6: グローバル環境で make-withdrawを定義した結果

作成します。ここまではこれまで⾒た例と同じですが、⼿続きの本体がそれ⾃
⾝ λ式であるというところが違います。
計算の⾯⽩いところは、⼿続き make-withdrawを引数に適⽤するときに起

こります。
(define W1 (make-withdraw 100))

まず、いつも通り環境 E1を⽴ち上げ、その中で仮引数 balanceを引数の 100
に束縛します。この環境の中で、make-withdraw の本体、つまり λ 式を評価
します。すると、新しい⼿続きオブジェクトが構築され、そのコード部分は
lambdaで規定されるものとなり、環境部分は、lambdaが評価された環境であ
る E1になります。結果としてできる⼿続きオブジェクトは、make-withdraw
の呼び出しに対して返される値になります。これがグローバル環境で W1に束
縛されます。これは、define⾃⾝がグローバル環境で評価されているからで
す。図 3.7に最終的な環境構造を⽰します。
これで、W1が引数に適⽤されるとどうなるか解析できるようになります。

(W1 50)
50

まずフレームを構築し、その中で W1の仮引数 amountを引数 50に束縛しま
す。注意すべき重要な点は、このフレームが持っている外側の環境が、グロー

E1

make-withdraw:

W1:

global
env

balance: 100

parameters:  balance

body:  ...
parameters:  amount

body:  (if (>= balance amount)
         (begin (set! balance (- balance amount))
                balance)
         "insufficient funds")

図 3.7: (define W1 (make-withdraw 100))を評価した結果

バル環境ではなく環境 E1だというところです。これは、W1⼿続きオブジェク
トによって規定されている環境が E1だからです。この新しい環境の中で、⼿
続きの本体を評価します。
(if (>= balance amount)

(begin (set! balance (- balance amount))
balance)

"Insufficient funds")

結果の環境構造を図 3.8に⽰します。評価されている式は、amountと balance
の両⽅を参照しています。amountは環境の最初のフレームで⾒つかり、balance
は外側の環境のポインタをたどることによって E1で⾒つかります。

set!が実⾏されると、E1中の balanceの束縛が変更されます。W1の呼び
出しが終わるときには balanceは 50になっていて、balanceを含むフレーム
はそれまで通り⼿続きオブジェクト W1によって指されています。balanceを
変更するコードを実⾏した、amountを束縛しているフレームは、もういらなく
なりました。このフレームを構築することになった⼿続きの呼び出しは終了し
ていますし、環境のほかの部分からこのフレームを指しているところもないか

E1

make-withdraw: ...

W1:

global
env

balance: 100

parameters:  amount

body:  ...

amount: 50

Here is the balance
that will be changed
by the set!

(if (>= balance amount)
    (begin (set! balance 
                 (- balance amount))
           balance)
    "insufficient funds")

図 3.8: ⼿続きオブジェクト W1を適⽤することによって作
成される環境

らです。次に W1が呼ばれるときには、新しいフレームが構築されて amountを
束縛し、その外側の環境は E1になります。⼿続きオブジェクト W1の局所状態
を持っておく “場所” としての役割を E1が果たしているということがわかり
ます。図 3.9に、W1を呼び出した後の様⼦を⽰します。
もう⼀度 make-withdrawを呼び出して、⼆つ⽬の “withdraw”オブジェク

トを作るとどうなるか⾒てみましょう。
(define W2 (make-withdraw 100))

こうすると、図 3.10のような環境構造が⽣成されます。この図は、W2が⼿
続きオブジェクトであるということ、つまり何らかのコードとひとつの環境の
ペアであるということを⽰しています。W2の環境 E2は、make-withdrawを呼
び出すことによって作られます。この環境は、balanceに対する局所的な束縛
を持つフレームを含んでいます。⼀⽅、W1と W2のコードは同じです。どちら
も、make-withdraw本体の λ式によって規定されるコードです。15 これで、な

15W1と W2はコンピュータ内の同じ物理コードを共有しているのか、それともそれぞ

E1

make-withdraw: ...

W1:

global
env

balance: 50

parameters:  amount

body:  ...

図 3.9: W1呼び出し後の環境

ぜ W1と W2が独⽴したオブジェクトとしてふるまうのかがわかりました。W1の
呼び出しは E1に格納された状態変数 balanceを参照し、W2の呼び出しは E2
に格納された balanceを参照しています。このため、⼀⽅のオブジェクトの局
所状態に変更を加えても、もう⼀⽅のオブジェクトには影響がありません。

練習問題 3.10: make-withdraw⼿続きでは、局所変数 balanceは
make-withdrawの仮引数として作られる。局所状態変数は、letを
使って、以下のように明⽰的に作ることもできる。
(define (make-withdraw initial-amount)

(let ((balance initial-amount))
(lambda (amount)

(if (>= balance amount)
(begin (set! balance (- balance amount))

balance)
"Insufficient funds"))))

letは⼿続き呼び出しに対する単純なシンタックスシュガーだと
いう1.3.2節の内容を思い出そう。

れコードのコピーを持っているのかということは、実装上の詳細に属することです。第
4章で実装するインタプリタでは、コードは実際には共有されることになります。

E1

W2:

W1:

global
env

balance: 50

parameters:  amount

body:  ...

E2 balance: 100

make-withdraw: ...

図 3.10: (define W2 (make-withdraw 100))を使って⼆つ
⽬のオブジェクトを作成

(let ((⟨var⟩ ⟨exp⟩)) ⟨body⟩)

上記の式は、次の式の別の書き⽅として解釈される。
((lambda (⟨var⟩) ⟨body⟩) ⟨exp⟩)

環境モデルを使って、この make-withdrawの別のバージョンを解
析し、上のような図を描いて次のやりとりを説明せよ。
(define W1 (make-withdraw 100))
(W1 50)
(define W2 (make-withdraw 100))

make-withdrawのどちらのバージョンも同じふるまいを持つオブ
ジェクトを作るということを⽰せ。環境構造は⼆つのバージョン
の間でどのように異なるだろうか。

3.2.4 内部定義
1.1.8節では、⼿続きが内部定義を持つことができるという考えを導⼊し、それ
によって、次の平⽅根を計算する⼿続きに出てくるようなブロック構造が作れ
るようになりました。
(define (sqrt x)

(define (good-enough? guess)
(< (abs (- (square guess) x)) 0.001))

(define (improve guess)
(average guess (/ x guess)))

(define (sqrt-iter guess)
(if (good-enough? guess)

guess
(sqrt-iter (improve guess))))

(sqrt-iter 1.0))

今回は、これらの内部定義がどうして望み通りにふるまうのかを⾒るのに環
境モデルを使うことができます。図 3.11は、式 (sqrt 2) を評価する途中の、
guessを 1として内部⼿続き good-enough?に対する初回の呼び出しを⾏った
時点の様⼦を⽰しています。
環境構造をよく⾒てください。sqrtはグローバル環境の中にある記号で、

⼿続きオブジェクトに束縛されています。その⼿続きオブジェクトに関連づけ
られている環境はグローバル環境です。sqrtが呼ばれると、グローバル環境の
下に新しい環境 E1が作られます。その中では、仮引数 xが 2に束縛されてい
ます。それから、sqrtの本体が E1の中で評価されます。sqrtの本体の最初の
式は以下の通りなので、
(define (good-enough? guess)

(< (abs (- (square guess) x)) 0.001))

この式を評価するときに、⼿続き good-enough?が環境 E1の中に定義されま
す。より正確に⾔うと、記号 good-enough?が E1の最初のフレームに追加さ
れ、環境 E1に関連づけられた⼿続きオブジェクトに束縛されるということに
なります。同じように、improveと sqrt-iterが E1の中で⼿続きとして定義
されます。簡単にするために、図 3.11では good-enough?に対する⼿続きオブ
ジェクトだけを⽰しています。
局所⼿続きが定義された後に、式 (sqrt-iter 1.0)が評価されます。この

式の評価が⾏われる環境は、やはり E1です。そのため、E1の中で sqrt-iter

268



parameters: x

body: (define good-enough? ...)
     (define improve ...)
     (define sqrt-iter ...)
     (sqrt-iter 1.0)

global
env

sqrt:

E1

x:2
good-enough?:
improve: ...
sqrt-iter: ...

parameters: guess

body: (< (abs ...)
         ...)

guess: 1

guess: 1

call to sqrt-iter

E2

call to good-enough?

E3

図 3.11: 内部定義を持つ sqrt⼿続き

に束縛されている⼿続きオブジェクトが、1を引数として呼び出されます。この
呼び出しで環境 E2が作られ、その中では sqrt-iterの仮引数 guessが 1に束
縛されています。sqrt-iterは、(E2の)guessの値を引数として good-enough?
を呼び出します。こうすると別の環境 E3が構築され、guess(good-enough?の
引数)は 1に束縛されます。sqrt-iterと good-enough?はどちらも guessと
いう名前の仮引数を持っていますが、これらは別々のフレームにある別々の局
所変数です。また、E2と E3は両⽅とも外側の環境として E1を持っています。
⼿続き sqrt-iterと good-enough?はどちらも環境部分として E1を持ってい
るためです。その結果、good-enough?の本体の中の記号 xは、E1に出てくる
xの束縛を参照することになります。これは、元の sqrt⼿続きの呼び出しに使
われた xの値ということになります。
このように、環境モデルを使うと、プログラムをモジュール化する便利な

テクニックとして局所⼿続きの定義を使ううえでの⼆つのキーとなる性質につ
いて説明できます。

• 局所⼿続きの名前は、それらを囲む⼿続きの外側の名前と衝突しない。
局所⼿続きの名前は、グローバル環境に束縛されるのではなく、⼿続き
が実⾏されるときに作られるフレームに束縛されるからである。

• 局所⼿続きは、単純に仮引数の名前を⾃由変数として使うだけで、それ
らを囲む⼿続きの引数にアクセスできる。これは、局所⼿続きの本体が、
外側の⼿続きの評価環境に属する環境で評価されるからである。
練習問題 3.11: 3.2.3節では、局所状態を持つ⼿続きのふるまいが
どのように環境モデルによって説明されるかを⾒た。ここでは、内
部定義がどのように動くのかを⾒た。典型的なメッセージパッシ
ング⼿続きは、これら両⽅の側⾯を持っている。3.1.1節の銀⾏⼝
座⼿続きについて考えてみよう。
(define (make-account balance)

(define (withdraw amount)
(if (>= balance amount)

(begin (set! balance (- balance amount))
balance)

"Insufficient funds"))
(define (deposit amount)

(set! balance (+ balance amount))
balance)

(define (dispatch m)
(cond ((eq? m 'withdraw) withdraw)

((eq? m 'deposit) deposit)
(else
(error "Unknown request: MAKE-ACCOUNT"

m))))
dispatch)

次の対話によって⽣成される環境構造を⽰せ。
(define acc (make-account 50))
((acc 'deposit) 40)
90

((acc 'withdraw) 60)
30

accの局所状態はどこに保存されているだろうか。次のように、別
の⼝座を定義したとする。
(define acc2 (make-account 100))

⼆つの⼝座の局所状態は、どのようにして別々に保存されること
になるだろうか。accと acc2では、環境構造のどの部分が共有さ
れているだろうか。

3.3 可変データによるモデル化
第 2章では、複合データを複数の部品からなる計算オブジェクトを作る⼿段と
して使う⽅法について取り上げました。これは、複数の側⾯を持つ現実世界の
オブジェクトをモデル化するためのものでした。また、第 2章ではデータ抽象
化の規律を導⼊しました。その規律に従うと、データ構造は、データオブジェ
クトを作るコンストラクタと、複合データオブジェクトの部品にアクセスする
セレクタによって規定することになっていました。しかし今では、データとい
うものには第 2章では触れなかったような別の側⾯があるということがわかっ
ています。状態が変化するオブジェクトからなるシステムをモデル化しようと
すると、複合データを構築したり選択を⾏ったりするだけでなく、それらを修
正する必要も出てきます。状態の変化する複合オブジェクトをモデル化するた
めに、ここでセレクタとコンストラクタに加えて、データオブジェクトを変更
するミューテータ (mutator)と呼ばれる演算を含めたデータ抽象化を設計する
ことにします。例えば、銀⾏システムをモデル化しようとすると、⼝座残⾼を
変更する必要があります。このため、銀⾏⼝座を表すデータ構造は、例えば次
のように指定された⼝座の残⾼を指定された新しい値に変更する演算ができる
ようにすることが考えられます。
(set-balance! ⟨account⟩ ⟨new-value⟩)

ミューテータが定義されているデータオブジェクトは、可変データオブジェク
ト (mutable data object)と呼ばれます。
第 2章では、複合データを合成するための汎⽤的な “糊”としてペアを導⼊

しました。この節では、ペアのための基本的なミューテータを定義することか
ら始め、可変データオブジェクトを構築するためのブロックとしてペアを使う

ことができるようにします。これらのミューテータはペアの表現⼒を⼤幅に拡
張し、それによって2.2節で取り組んだ列や⽊構造以外のデータ構造も作れる
ようになります。また、シミュレーションの例もいくつか取り上げます。それ
らの例では、複雑なシステムを局所状態を持つオブジェクトの集合としてモデ
ル化することになります。

3.3.1 可変リスト構造
ペアに対する基本演算—cons, car, cdr—は、リスト構造を構築し、またリス
ト構造から部品を選択するために使うことができます。しかし、これらはリス
ト構造を変更することはできません。これまで学んできた、appendや listの
ようなリスト演算についても同じことが⾔えます。これらは、cons, car, cdr
によって定義できるものだからです。リスト構造を変更するには、新しい命令
が必要となります。
ペアに対する基本ミューテータは set-car! と set-cdr! です。set-car!

は⼆つの引数を取り、⼀つ⽬の引数はペアでなければなりません。set-car!は
このペアを変更し、carポインタを set-car!の第⼆引数へのポインタに置き
換えます。16

例として、図 3.12に⽰すように、xがリスト ((a b) c d)に束縛され、yが
リスト (e f)に束縛されているという場合について考えます。式 (set-car! x
y)を評価すると、xが束縛されているペアが変更され、その carを yの値で置
き換えます。演算の結果を図 3.13に⽰します。構造 xは変更され、今では ((e
f) c d)と表⽰されるような構造になっています。置き換えられたポインタが
指していたリスト (a b)を表す⼆つのペアは、今では元の構造から切り離され
ています。17

図 3.13と図 3.14を⾒⽐べてみてください。後者は、xと yが 図 3.12の元の
リストに束縛されている状態で (define z (cons y (cdr x)))を実⾏した結
果を⽰したものです。変数 zは cons演算によって作られた新しいペアに束縛
されるようになっていますが、xが束縛されているリストは変化していません。

16set-car!と set-cdr!の返す値は実装依存です。set!同様、これらの演算はそれら
の持つ効果のために使うものです。

17ここから、リストに対して変更演算を⾏うと、アクセス可能な構造の部品になって
いない “garbage” (ゴミ)ができるかもしれないということがわかります。5.3.2節では、
Lispのメモリ管理システムはガベージコレクタ (garbage collector)というものを含んで
いて、それによって必要のないペアが使っていたメモリ空間をリサイクルするというこ
とを学びます。

x

e f

a b

図 3.12: リスト x: ((a b) c d)と y: (e f)

c d

y

x

e f

a b

図 3.13: 図 3.12のリストに対する (set-car! x y)の効果

図 3.14: 図 3.12のリストに対する (define z (cons y (cdr
x)))の効果

set-cdr!演算は set-car!と似ています。唯⼀の違いは、置き換えるのがペア
の carポインタではなく cdrポインタだということです。図 3.12のリストに対
して (set-cdr! x y)を実⾏したときの効果を図 3.15に⽰します。ここでは、x
の cdrポインタが (e f)に対するポインタに置き換えられています。また、x
の cdrだったリスト (c d)は、今では構造から取り外されています。

consは新しいペアを作ることによって新しいリスト構造を組み⽴てるもの
ですが、set-car!と set-cdr!は既存のペアを修正するものです。実際のとこ
ろ、これら⼆つのミューテータと、既存のリスト構造の⼀部でない新しいペア
を返す get-new-pairという⼿続きがあれば、それらによって consを実装する
こともできます。まず新しいペアを取得し、指定されたオブジェクトを carと
cdrポインタに設定し、その新しいペアを consの結果として返します。18

(define (cons x y)
(let ((new (get-new-pair)))

(set-car! new x)
(set-cdr! new y)
new))

18get-new-pairは、Lispの実装に必要となるメモリ管理の⼀部として実装する必要が
ある演算のひとつです。これについては 5.3.1節で検討します。

図 3.15: 図 3.12のリストに対する (set-cdr! x y)の効果

練習問題 3.12:
2.2.1節では、次のリスト追加⼿続きを紹介した。
(define (append x y)

(if (null? x)
y
(cons (car x) (append (cdr x) y))))

appendは、xの要素を連続で yに consしていくことによって新し
いリストを作る。⼿続き append!は appendと似ているが、コンス
トラクタではなくミューテータである。これは xの最後のペアを
修正し、cdrが yになるようにして⼆つのリストをつなぎ合わせ
ることによって追加を⾏う (空の xを引数として append!を呼ぶ
とエラーになる)。
(define (append! x y)

(set-cdr! (last-pair x) y)
x)

ここで、last-pairは引数の最後のペアを返す⼿続きである。
(define (last-pair x)

(if (null? (cdr x)) x (last-pair (cdr x))))

次のやりとりについて考えよ。
(define x (list 'a 'b))
(define y (list 'c 'd))
(define z (append x y))
z
(a b c d)
(cdr x)
⟨response⟩
(define w (append! x y))
w
(a b c d)
(cdr x)
⟨response⟩

空欄の ⟨response⟩はどのようになるだろうか。箱とポインタの図
を描き、答えについて説明せよ。

練習問題 3.13: 次の make-cycle⼿続きについて考えよ。これは、
練習問題 3.12で定義した last-pair⼿続きを使っている。
(define (make-cycle x)

(set-cdr! (last-pair x) x)
x)

次の式によって作られる構造 zを表す箱とポインタの図を描け。
(define z (make-cycle (list 'a 'b 'c)))

(last-pair z)を計算しようとすると、どうなるだろうか。
練習問題 3.14: 次の⼿続きはわかりにくいが、とても役に⽴つ。
(define (mystery x)

(define (loop x y)
(if (null? x)

y
(let ((temp (cdr x)))

(set-cdr! x y)
(loop temp x))))

(loop x '()))

loop は “⼀時的” な変数 temp を使って、x の cdr の古い値を保
持する。これは、次の⾏の set-cdr!が cdrを破壊するためであ
る。⼀般的に mysteryが何をするのか説明せよ。vは (define v
(list 'a 'b 'c 'd))によって定義されたものだとする。vが束
縛されるリストを表す箱とポインタの図を描け。次に、(define w
(mystery v))を評価したとする。この式を評価した後の vと wの
構造を表す箱とポインタの図を描け。vと wの値として何が表⽰さ
れるだろうか。

共有とアイデンティティ
3.1.3節では、代⼊を導⼊することによって出てくる “同⼀性”と “変更”とい
う理論的な問題について触れました。これらの問題は、個々のペアが別々のデ
ータオブジェクトの間で 共有 (shared)されると実際に起こるようになります。
例えば、以下のように作られる構造について考えてみましょう。

z1

x

a b

図 3.16: (cons x x)によって作られるリスト z1

(define x (list 'a 'b))
(define z1 (cons x x))

図 3.16に⽰したように、z1は、carと cdrが両⽅とも xという同じペアを指し
ています。このように、z1の carと cdrによって xが共有できるようになって
いるのは、consが単純な⽅法で実装されているからです。⼀般的に、consを
使ってリストを構築すると、多くの個々のペアが多くの別々の構造によって共
有される、ペアの絡まり合った構造ができます。
図 3.16に対して、図 3.17は次の式によって作られる構造を⽰しています。

(define z2 (cons (list 'a 'b) (list 'a 'b)))

この構造の中では、⼆つの (a b)リストに含まれるペアは、実際の記号は共有
されていても、区別可能です。19

リストとして考えると、z1と z2は両⽅とも “同じ”リスト、つまり ((a b)
a b)を表現しています。⼀般的に、リストに対して cons, car, cdrだけを使っ
て操作するのであれば、共有は完全に検出不可能です。しかし、リスト構造に
対して変更を許可するとなると、共有は意味を持つようになります。共有によ
って違いが出てくる例として、次の⼿続きについて考えてみましょう。これは、
適⽤対象の構造の carを修正するものです。

19consを呼び出すたびに新しいペアが返されるため、⼆つのペアは区別可能です。記
号は共有されています。Schemeでは、どんな名前に対しても⼀意の記号が存在します。
Schemeは記号を変更する⼿段を提供していないため、この共有を検出することはできま
せん。また、そのように記号が共有されていることによって、単純にポインタが同じか
どうかをチェックするだけの eq?によって記号の⽐較ができるようになっているという
ことにも注意してください。

a b

z2

図 3.17: (cons (list 'a 'b) (list 'a 'b))によって作ら
れるリスト z2

(define (set-to-wow! x) (set-car! (car x) 'wow) x)

z1と z2は “同じ”構造のはずなのに、これらに set-to-wow!を適⽤すると、異
なる結果が返ってきます。z1の場合は、carと cdrは同じペアなので、carを
変更すると cdrも変更することになります。z2の場合は、carと cdrは別のも
のなので、set-to-wow!は carしか変更しません。
z1
((a b) a b)
(set-to-wow! z1)
((wow b) wow b)
z2
((a b) a b)
(set-to-wow! z2)
((wow b) a b)

リスト構造の中にある共有を検出する⽅法として、述語 eq?を使うというも
のがあります。この述語は、⼆つの記号が同じかどうかテストする⽅法とし
て2.3.1節で紹介したものです。より⼀般的に、(eq? x y)は xと yが同じオブ
ジェクトかどうか (つまり、xと yがポインタとして等しいかどうか)をテスト
します。そのため、図 3.16と 図 3.17のように定義された z1と z2では、(eq?
(car z1) (cdr z1)) は真になり、(eq? (car z2) (cdr z2))は偽になります。
以降の節で⾒ていくように、共有を活⽤すると、ペアで表現可能なデータ

構造の範囲を⼤きく広げることができます。⼀⽅、共有は危険なものになる可
能性もあります。ある構造に加えた変更が、その変更部分をたまたま共有して
いる別の構造にも影響することになるからです。変更演算の set-car!と set-

cdr!を使う際には注意する必要があります。扱っているデータオブジェクト
についてよく理解していないと、変更が予期しない結果を招くことがあるかも
しれないからです。20

練習問題 3.15: 箱とポインタの図を描いて、上記の構造 z1と z2
に対する set-to-wow!の効果を説明せよ。
練習問題 3.16: Ben Bitdiddleは任意のリスト構造内のペアの数を
数える⼿続きを書くことにした。“簡単だよな”と彼は考えた。“ど
んな構造でも、carの分の数と cdrの分の数を⾜して、それにその
ペア⾃⾝を数えるために 1を⾜せば、ペアの数が求められる。” そ
して、Benは次の⼿続きを書いた。
(define (count-pairs x)

(if (not (pair? x))
0
(+ (count-pairs (car x))

(count-pairs (cdr x))
1)))

この⼿続きが正しくないということを⽰せ。具体的には、ちょう
ど三つのペアからなるリスト構造で、Benの⼿続きが 3, 4, 7を返
すもの、また⼿続きが終了しないものを考え、それらを表す箱と
ポインタの図を描け。
練習問題 3.17: 練習問題 3.16の count-pairs⼿続きの正しいバー
ジョンを考え、任意の構造について区別可能なペアの数を返すよ
うにせよ。(ヒント：カウント済みのペアを記録しておく補助デー
タ構造を維持しながら構造をたどれ)。

20可変データオブジェクトの共有を扱うことの難しさには、3.1.3節で提起した “同⼀
性”と “変更”の問題が根底にあります。そこでは、私たちの⾔語で変更を許すようにす
るためには複合データは構成部品の集まりとは別の “アイデンティティ”を持たなければ
ならないということに触れました。Lispでは、この “アイデンティティ”は、eq?によっ
てテストされるような性質だと考えます。つまり、ポインタが同じかどうかということ
になります。多くの Lisp実装では、ポインタの実装は本質的にはメモリアドレスなので、
データオブジェクト “そのもの”は特定のメモリ上の場所のセットに格納された情報であ
ると取り決めることによって、オブジェクトのアイデンティティを定義するという “問
題を解決”しています。これは単純な Lispのプログラムには⼗分ですが、計算モデルの
“同⼀性”という問題を解く⼀般的な⽅法とはとても⾔えないものです。

練習問題 3.18: リストを検査し、循環を持つかどうか、つまり連続
して cdrを取ることによってリストの終端を探そうとするプログ
ラムが無限ループになるかどうかを判定する⼿続きを書け。練習
問題 3.13でそのようなリストを構築した。
練習問題 3.19: 定数量の空間しか使わないアルゴリズムによっ
て練習問題 3.18をやり直せ (これにはとても巧妙なアイデアが必
要となる)。

変更は代⼊にすぎない
複合データを導⼊したとき、2.1.3節ではペアが純粋に⼿続きだけによって表
現できるということを⾒ました。
(define (cons x y)

(define (dispatch m)
(cond ((eq? m 'car) x)

((eq? m 'cdr) y)
(else (error "Undefined operation: CONS" m))))

dispatch)
(define (car z) (z 'car))
(define (cdr z) (z 'cdr))

可変データについても同じことが⾔えます。可変データオブジェクトは、代
⼊と局所状態を使って⼿続きとして実装できます。例えば、3.1.1節で make-
accountを使って銀⾏⼝座を実装したのと似たようなやり⽅で、上記のペアの
実装を拡張して set-car!と set-cdr!を扱えるようにできます。
(define (cons x y)

(define (set-x! v) (set! x v))
(define (set-y! v) (set! y v))
(define (dispatch m)

(cond ((eq? m 'car) x)
((eq? m 'cdr) y)
((eq? m 'set-car!) set-x!)
((eq? m 'set-cdr!) set-y!)
(else
(error "Undefined operation: CONS" m))))

dispatch)

(define (car z) (z 'car))
(define (cdr z) (z 'cdr))
(define (set-car! z new-value)

((z 'set-car!) new-value) z)
(define (set-cdr! z new-value)

((z 'set-cdr!) new-value) z)

可変データのふるまいを説明するためには、理論的には、代⼊だけあれば⼗分
です。私たちの⾔語で set!を認めたとたんに、代⼊の問題だけでなく、可変
データ⼀般に関する問題がすべて引き起こされることになります。21

練習問題 3.20: 上記の⼿続きによるペアの実装を使って、次の式
の列を評価する。
(define x (cons 1 2))
(define z (cons x x))
(set-car! (cdr z) 17)
(car x)
17

評価の様⼦を、環境の図を描いて説明せよ (練習問題 3.11と⽐較せ
よ)。

3.3.2 キューの表現
ミューテータ set-car!、set-cdr!を使うと、cons, car, cdr だけでは構築で
きないようなデータがペアを使って構築できるようになります。この節では、
ペアを使ってキューと呼ばれるデータ構造を表現する⽅法を学びます。3.3.3
節では、テーブルと呼ばれるデータ構造を表現するやり⽅について学ぶことに
なります。
キュー (queue)とは、項⽬を⼀⽅の端 (キューの終端 (rear)) に挿⼊し、も

う⼀⽅の端 (先端 (front))から削除する列です。図 3.18は、最初は空だったキ
ューにアイテム aと bが挿⼊される様⼦を⽰しています。次に aが削除され、
cと dが挿⼊され、bが削除されています。項⽬は常に挿⼊した順番に削除さ

21⼀⽅、実装⾯から⾒ると、代⼊を実装するためには環境を変更することが必要となり
ますが、この環境というのはそれ⾃⾝が可変データ構造です。このように、代⼊と変更
は等価だということです。どちらも、もう⼀⽅によって実装できます。

Operation                 Resulting Queue

(define q (make-queue))
(insert-queue! q 'a)      a
(insert-queue! q 'b)      a b
(delete-queue! q)         b
(insert-queue! q 'c)      b c
(insert-queue! q 'd)      b c d
(delete-queue! q)         c d

図 3.18: キュー演算

れるため、キューは先⼊れ先出し (FIFO(first in, first out))バッファと呼ばれ
ることもあります。
データ抽象化を使うと、キューは以下の演算の集合によって定義されるも

のと考えることができます。
• コンストラクタ：(make-queue)は、空のキュー (項⽬を持たないキュー)
を返す。

• ⼆つのセレクタ：
(empty-queue? ⟨queue⟩) はキューが空かどうかテストする。
(front-queue ⟨queue⟩) はキューの先頭にあるオブジェクトを返す。キ
ューが空ならエラーを出す。キューは変更しない。

• ⼆つのミューテータ：
(insert-queue! ⟨queue⟩ ⟨item⟩) はキューの最後尾にアイテムを挿⼊
し、変更されたキューを値として返す。
(delete-queue! ⟨queue⟩) はキューの先頭のアイテムを削除し、変更さ
れたキューを値として返す。削除前にキューが空であればエラーを出す。

キューはアイテムの列なので、当然通常のリストとして表現できるところです。
キューの先頭はリストの carとなり、キューへの項⽬の挿⼊はリストの終端に
新しい要素を追加することとなり、キューからのアイテムの削除は単にリスト
の cdrを取るということになるでしょう。しかし、この表現は⾮効率的です。
というのは、項⽬を挿⼊するためにはリストを終端まで⾛査しなければいけな
いからです。リストを⾛査するための⼿段は cdr演算を連続して使うことだけ
なので、この⾛査は n項⽬のリストに対して Θ(n)ステップが必要になります。

q

a b

rear-ptr

図 3.19: 先端と終端のポインタを持つリストとしてのキュ
ーの実装

リスト表現に単純な変更を加えるだけで、必要なステップが Θ(1)、つまり必要
なステップ数がキューの⻑さと独⽴なキュー演算の実装ができるようになり、
この⽋点を克服できます。
リスト表現の難しいところは、リストの終端を⾒つけるためにリストを⾛

査しないといけないということです。なぜ⾛査しないといけないかというと、
リストをペアの鎖として表現する標準的なやり⽅では、先頭へのポインタはす
ぐに得られるのに対し、リストの終端へのポインタは簡単にアクセスできない
ようになっているからです。その⽋点を避けるためには、キューをリストとし
て表現したうえで、リストの最終ペアを指す追加のポインタを加えるように修
正します。こうすると、項⽬を挿⼊する際に終端ポインタを⾒ることによって
リストの⾛査を避けられます。
このやり⽅では、キューは front-ptr, rear-ptrというポインタのペアと

して表現できます。ポインタはそれぞれ通常のリストの先頭と最後のペアを指
しています。キューを識別可能なオブジェクトにするために、consを使って⼆
つのポインタを結合します。つまり、キュー⾃⾝は⼆つのポインタの consに
なります。図 3.19はこの表現を図⽰したものです。
キュー演算を定義するには以下の⼿続きを使います。これらは、キューの

先端・終端ポインタを選択・変更できるようにするものです。
(define (front-ptr queue) (car queue))
(define (rear-ptr queue) (cdr queue))
(define (set-front-ptr! queue item)

(set-car! queue item))

(define (set-rear-ptr! queue item)
(set-cdr! queue item))

これで、実際のキュー演算を実装できるようになります。先端のポインタが空
リストであればキューは空であるとします。
(define (empty-queue? queue)

(null? (front-ptr queue)))

make-queueコンストラクタは、初期状態の空キューとして、carと cdrの両⽅
が空リストであるペアを返します。
(define (make-queue) (cons '() '()))

キューの先頭の項⽬を選択するには、先端ポインタが指すペアの carを返し
ます。
(define (front-queue queue)

(if (empty-queue? queue)
(error "FRONT called with an empty queue" queue)
(car (front-ptr queue))))

キューにアイテムを挿⼊するには、結果として図 3.20のようになるような⼿順
を実⾏します。まず、carが挿⼊する項⽬で cdrが空リストという新しいペア
を作ります。もしキューが元々空だったら、先端と終端のポインタが両⽅とも
この新しいペアを指すようにします。そうでなければ、キューの最終ペアが新
しいペアを指すように変更し、また終端ポインタも新しいペアを指すようにし
ます。
(define (insert-queue! queue item)

(let ((new-pair (cons item '())))
(cond ((empty-queue? queue)

(set-front-ptr! queue new-pair)
(set-rear-ptr! queue new-pair)
queue)

(else
(set-cdr! (rear-ptr queue) new-pair)
(set-rear-ptr! queue new-pair)
queue))))

図 3.20: 図 3.19のキューに (insert-queue! q 'd) を使っ
た結果

キューの先頭の項⽬を削除するのは、先頭ポインタがキューの⼆番⽬の項
⽬を指すようにするだけで⼤丈夫です (図 3.21参照)。22

(define (delete-queue! queue)
(cond ((empty-queue? queue)

(error "DELETE! called with an empty queue" queue))
(else (set-front-ptr! queue (cdr (front-ptr queue)))

queue)))

練習問題 3.21: Ben Bitdiddleは、上記のキューの実装をテストす
ることにした。彼は Lispインタプリタにキューの⼿続きを⼊⼒し、
それから試しに使ってみた。
(define q1 (make-queue))
(insert-queue! q1 'a)
((a) a)
(insert-queue! q1 'b)
((a b) b)
(delete-queue! q1)
((b) b)

22最初の項⽬がキューの最終項⽬でもある場合、削除後に先端ポインタは空リストにな
り、これがキューが空だという印になります。終端ポインタは依然として削除済みの項
⽬を指し続けることになりますが、わざわざ更新する必要はありません。empty-queue?
は先端ポインタしか⾒ないからです。

図 3.21: 図 3.20のキューに (delete-queue! q) を使っ
た結果

(delete-queue! q1)
(() b)

“全然ダメじゃないか”と彼は⽂句を⾔う。“インタプリタの応答を
⾒てみろよ。最後の項⽬がキューに⼆回挿⼊されているじゃない
か。それに、項⽬を両⽅とも削除しても、⼆つ⽬の bは残ったまま
だ。つまり、キューは空になっていないってことだ。空になってい
ないといけないのに” Eva Lu Atorは、Benが挙動を誤解している
と⾔う。“項⽬がキューに⼆回⼊っているわけじゃなくて” 彼⼥は
説明する。“Lisp標準の表⽰プログラムが、キューの表現をどう解
釈していいかわからないってだけ。キューをちゃんと表⽰させた
いなら、キューを表⽰する⼿続きを⾃分で書かないと” Eva Luの
話について解説せよ。具体的には、なぜ Benの例がそのような表
⽰結果を出⼒するのか説明せよ。キューを⼊⼒として取り、キュ
ー内の項⽬列を表⽰する⼿続き print-queueを定義せよ。
練習問題 3.22: キューは、ポインタのペアとして表現する以外に、
局所状態を持つ⼿続きとして構築することもできる。局所状態は
通常のリストの先端と終端へのポインタからなる。したがって、⼿
続き make-queueは以下のような形になる。
(define (make-queue)

(let ((front-ptr . . . )
(rear-ptr . . . ))

⟨definitions of internal procedures⟩
(define (dispatch m) . . .)
dispatch))

make-queueの定義を完成させ、この表現によるキューの演算を実
装せよ。
練習問題 3.23: 両端キュー (deque)(“double-ended queue”)は、先
端と終端のどちらに対しても項⽬の挿⼊と削除が⾏える列で
ある。deque に対する演算は、コンストラクタ make-deque、述
語 empty-deque?、セレクタ front-deque、rear-deque、ミューテ
ータ front-insert-deque!,rear-insert-deque!, front-delete-
deque!, rear-delete-deque!である。ペアによって dequeを表現
するやり⽅を⽰せ。また、演算を実装せよ。23 すべての演算はΘ(1)
ステップで完了しなければならない。

3.3.3 テーブルの表現
第 2章で集合を表現するいろいろなやり⽅について学んだ際に、2.3.3節では、
識別⽤キーによって索引づけされたレコードのテーブルを保持するというタス
クについて触れました。2.4.3節のデータ主導プログラミングを実装するにあ
たっては、⼆つのキーによって情報を格納したり取り出したりする⼆次元のテ
ーブルが⼤活躍しました。ここでは、可変リスト構造としてテーブルを構築す
るやり⽅について⾒ていきます。
まず、⼀次元の表について考えます。⼀次元の表では、それぞれの値は単独

のキーの下に格納されます。このテーブルをレコードの列として実装し、それ
ぞれのレコードは、キーとそれに関連づけられた値とのペアとして実装します。
レコードはペアによって連結して列状にします。連結⽤ペアの carは連続した
レコードを指しています。これらの連結⽤ペアは、テーブルの背⾻ (backbone)
と⾔います。テーブルに新しいレコードを追加するときに変更できる場所が
必要なので、テーブルは頭つきリスト (headed list)として構築します。頭つき
リストは、先頭に特別な背⾻のペアを持っています。これはダミーの “レコー
ド”—今回の場合、適当に選んだ記号 *table*—を持っています。図 3.22は、次
のテーブルに対する箱とポインタの図を⽰したものです。

23インタプリタに循環を含む構造を表⽰させないように注意すること (練習問題 3.13参
照)。

289



a b c1 2 3

*table*

table

図 3.22: 頭つきリストとして表現されたテーブル

a: 1
b: 2
c: 3

テーブルから情報を取り出すには lookupという⼿続きを使います。この⼿続
きは引数としてキーを取り、関連づけられた値 (キーの下に格納された値がな
ければ偽)を返します。lookupは、引数としてキーとレコード列を取る assoc
演算によって定義します。assocはダミーレコードを⾒ないことに注意してく
ださい。assocは、与えられたキーを carとして持つレコードを返します。24

それから lookupは、assocに返されたレコードが偽でないことをチェックし、
そのレコードの値 (cdr)を返します。
(define (lookup key table)

(let ((record (assoc key (cdr table))))
(if record

(cdr record)
false)))

(define (assoc key records)
(cond ((null? records) false)

((equal? key (caar records)) (car records))
(else (assoc key (cdr records)))))

24assocは equal?を使うため、記号・数値・リスト構造のキーを認識できます。

特定のキーの下に値を挿⼊するには、まず assocを使って、すでにそのキーを
持つレコードがテーブルにないか確認します。なければ、キーと値を consして
新しいレコードを作り、それをテーブルのレコード列の先頭、つまりダミーレ
コードの後ろに挿⼊します。すでにそのキーのレコードがあるなら、そのレコ
ードの cdrを指定された新しい値に設定します。テーブルのヘッダは、新しい
レコードを挿⼊する際に修正する固定の場所という役割を果たしています。25

(define (insert! key value table)
(let ((record (assoc key (cdr table))))

(if record
(set-cdr! record value)
(set-cdr! table

(cons (cons key value)
(cdr table)))))

'ok)

新しいテーブルを構築するのは、記号 *table*を持つリストを作るだけで終わ
りです。
(define (make-table)

(list '*table*))

⼆次元テーブル
⼆次元テーブルでは、それぞれの値は⼆つのキーによって索引づけがされてい
ます。そのようなテーブルは、⼀次元テーブルのそれぞれのキーがサブテーブ
ルを特定するようにすることで作れます。図 3.23は、次のテーブルに対する箱
とポインタの図を⽰したものです。

math: +: 43 letters: a: 97
-: 45 b: 98
*: 42

25このため、最初の背⾻ペアは、テーブル “そのもの”を表現するオブジェクトという
ことになります。つまり、テーブルを指すポインタというのは、このペアを指すポイン
タだということです。この同じ背⾻ペアは、常にテーブルのスタート地点になっていま
す。もしこういうやり⽅をしていなければ、insert!が新しいレコードを追加する際に、
テーブルの新しいスタート地点の値を返さなければならなくなるところです。

このテーブルは、⼆つのサブテーブルを持っています。(サブテーブルには特別
なヘッダ記号は必要ありません。サブテーブルを特定するキーがその⽬的を果
たすためです)。
項⽬を探すときには、⼀つ⽬のキーを使って正しいサブテーブルを特定し

ます。次に、⼆つ⽬のキーを使ってサブテーブル中のレコードを特定します。
(define (lookup key-1 key-2 table)

(let ((subtable
(assoc key-1 (cdr table))))

(if subtable
(let ((record

(assoc key-2 (cdr subtable))))
(if record

(cdr record)
false))

false)))

キーのペアの下に新しい項⽬を挿⼊するには、assocを使って⼀つ⽬のキ
ーの下にサブテーブルが格納されているかどうか確認します。もしなければ、
ひとつのレコード (key-2, value)を持つ新しいサブテーブルを構築し、それを
テーブルの⼀つ⽬のキーの下に挿⼊します。もし⼀つ⽬のキーに対するサブテ
ーブルがすでに存在するなら、上で述べた⼀次元テーブルの挿⼊法を使って、
そのサブテーブルに新しいレコードを挿⼊します。
(define (insert! key-1 key-2 value table)

(let ((subtable (assoc key-1 (cdr table))))
(if subtable

(let ((record (assoc key-2 (cdr subtable))))
(if record

(set-cdr! record value)
(set-cdr! subtable

(cons (cons key-2 value)
(cdr subtable)))))

(set-cdr! table
(cons (list key-1

(cons key-2 value))
(cdr table)))))

'ok)

292



+ - *43 45 42

*table*

a b97 98

letters

math

table

図 3.23: ⼆次元テーブル

局所テーブルの作成
上で定義した lookupと insert!演算は、引数としてテーブルを取ります。こ
のことによって、⼆つ以上のテーブルにアクセスするプログラムが使えるよう
になります。複数のテーブルを扱う別の⽅法としては、それぞれのテーブルに
対して別々の lookupと insert!⼿続きを持つというものがあります。これは、
内部テーブルを局所状態として保持するオブジェクトとしてテーブルを⼿続
き的に表現することによって達成できます。適切なメッセージが送られると、
この “テーブルオブジェクト”は内部テーブルに対する演算を⾏う⼿続きを提
供します。以下に、この⽅法で表現した⼆次元テーブル⽣成プログラムを⽰し
ます。

(define (make-table)
(let ((local-table (list '*table*)))

(define (lookup key-1 key-2)
(let ((subtable

(assoc key-1 (cdr local-table))))
(if subtable

(let ((record
(assoc key-2 (cdr subtable))))

(if record (cdr record) false))
false)))

(define (insert! key-1 key-2 value)
(let ((subtable

(assoc key-1 (cdr local-table))))
(if subtable

(let ((record
(assoc key-2 (cdr subtable))))

(if record
(set-cdr! record value)
(set-cdr! subtable

(cons (cons key-2 value)
(cdr subtable)))))

(set-cdr! local-table
(cons (list key-1 (cons key-2 value))

(cdr local-table)))))
'ok)

(define (dispatch m)
(cond ((eq? m 'lookup-proc) lookup)

((eq? m 'insert-proc!) insert!)
(else (error "Unknown operation: TABLE" m))))

dispatch))

make-table を使うと、2.4.3 節でデータ主導プログラミングに使った get と
putを次のように実装できます。
(define operation-table (make-table))
(define get (operation-table 'lookup-proc))
(define put (operation-table 'insert-proc!))

294



getは⼆つのキーを引数に取り、putは⼆つのキーと値を引数に取ります。ど
ちらの操作も、make-table呼び出し時に作られるオブジェクトにカプセル化
された同じ局所テーブルにアクセスします。

練習問題 3.24: 上のテーブル実装では、キーの等価性は (assocか
ら呼ばれる)equal?によってテストされている。これは常に適切な
テストというわけではない。例えば、数字をキーとするテーブル
を扱っている場合、検索する数字に対して完全に⼀致する必要が
なく、いくらかの誤差の範囲内のものであればいいということも
考えられる。キーの “等価性”をテストするのに使う⼿続き same-
key?を引数として取るテーブルコンストラクタ make-tableを定
義せよ。make-tableは dispatch⼿続きを返すようにして、それ
を使って局所テーブルに対する適切な lookup, insert!⼿続きに
アクセスできるようにすること。
練習問題 3.25:⼀次元と⼆次元のテーブルを⼀般化し、任意の数の
キーの下に値を格納するテーブルを実装するやり⽅を⽰せ。値に
よってキーの数は異なることもある。lookup, insert!⼿続きは、
テーブルにアクセスするためのキーのリストを⼊⼒として取る。
練習問題 3.26:上のように実装したテーブルを検索するには、レコ
ードのリストを⾛査する必要がある。これは、基本的に2.3.3節の
順序なしリスト表現である。⼤きなテーブルに対しては、別のや
り⽅でテーブルを構造化するほうが効率がよくなるかもしれない。
(キー, 値)のレコードを⼆分⽊を使ってまとめるようなテーブル
の実装について説明せよ。キーは、何らかの⽅法 (数の⼤⼩順やア
ルファベット順など)によって順序づけができるものとする (第 2
章の練習問題 2.66と⽐較せよ)。
練習問題 3.27: メモ化 (memoization)(テーブル化 (tabulation) と
も)とは、以前計算したことのある値を⼿続きが局所テーブルに記
録できるようにするテクニックである。このテクニックを使うこ
とでプログラムのパフォーマンスが⼤きく変わることもある。メ
モ化された⼿続きは、以前の呼び出しの値をテーブルに保持する。
キーにはその値を⽣成した引数を使う。メモ化された⼿続きが値
を計算するよう求められると、まずテーブルをチェックしてその
値が⼊っていないか⾒て、もし⼊っていればその値をそのまま返
す。⼊っていなければ、普通のやり⽅で新しい値を計算し、それを

テーブルに格納する。メモ化の例として、1.2.2節のフィボナッチ
数を計算する指数的プロセスを思い出そう。
(define (fib n)

(cond ((= n 0) 0)
((= n 1) 1)
(else (+ (fib (- n 1)) (fib (- n 2))))))

同じ⼿続きをメモ化したバージョンは次のようになる。
(define memo-fib

(memoize
(lambda (n)

(cond ((= n 0) 0)
((= n 1) 1)
(else (+ (memo-fib (- n 1))

(memo-fib (- n 2))))))))

メモ化⼿続きは以下のように定義される。
(define (memoize f)

(let ((table (make-table)))
(lambda (x)

(let ((previously-computed-result
(lookup x table)))

(or previously-computed-result
(let ((result (f x)))

(insert! x result table)
result))))))

環境の図を描き、(memo-fib 3)の計算を分析せよ。なぜ memo-fib
が n番⽬のフィボナッチ数を nに⽐例するステップ数で計算でき
るのか説明せよ。単に memo-fibを (memoize fib)と定義してい
たとしても、この構想はうまくいくだろうか。

3.3.4 デジタル回路シミュレータ
複雑なデジタルシステム (例えばコンピュータなど)の設計は重要な⼯学活動
です。デジタルシステムは、単純な要素を相互接続することで構築されます。

Inverter And-gate Or-gate

図 3.24: デジタル論理回路シミュレータにおける基本関数

個々の要素のふるまいは単純ですが、それらが構成するネットワークのふるま
いはとても複雑になることもあります。回路設計案のコンピュータシミュレー
ションは、デジタルシステムエンジニアに使われる重要なツールです。この節
では、デジタル論理シミュレーションを実⾏するシステムを設計します。この
システムは、イベント駆動シミュレーション (event-driven simulation)と呼ば
れる種類のプログラムの例となります。この種類のプログラムでは、アクショ
ン (“イベント”)は後になってさらなるイベントを引き起こし、それらのイベン
トがまたさらなるイベントを引き起こし . . .という繰り返しになっています。
本書での回路の計算モデルは、回路を構成する基本部品に対応するオブジ

ェクトによって構成されることになります。まず、デジタル信号 (digital signal)
を伝える 線 (wire)があります。デジタル信号は、ある特定の瞬間には、⼆つの
可能な値である 0と 1のどちらかしか取りません。ほかには、いろいろなタイ
プのデジタル 関数箱 (function box)もあり、⼊⼒信号を伝える回路と、ほかの
出⼒回路とを接続しています。これらの箱は、⼊⼒信号から計算される信号を
出⼒します。出⼒信号は、関数箱のタイプによって時間的な遅れが⽣じます。
例えば、インバータ (inverter)は⼊⼒を反転する基本関数箱です。インバータ
に対する⼊⼒信号が 0に変化すると、インバータ遅延時間 1単位後にインバー
タは出⼒信号を 1に変えます。インバータに対する⼊⼒信号が 1に変化すると、
インバータ遅延時間 1単位後にインバータは出⼒信号を 0に変えます。インバ
ータは図 3.24のような記号として表します。同じ図 3.24に⽰しているANDゲ
ート (and-gate)は、⼆つの⼊⼒とひとつの出⼒を持つ基本関数箱です。この関
数箱は、⼊⼒の論理積 (logical and)となる値に出⼒信号を駆動します。つまり、
⼊⼒信号が両⽅とも 1になると、ANDゲート遅延時間 1単位後に、ANDゲー
トは出⼒信号を 1にするということです。そうでなければ、出⼒は 0です。OR
ゲート (or-gate)は、似たような 2⼊⼒の基本関数箱で、⼊⼒の 論理和 (logical
or)となる値に出⼒を駆動します。つまり、少なくともひとつの⼊⼒信号が 1
であれば出⼒は 1になり、そうでなければ出⼒は 0になるということです。
基本関数を接続すると、より複雑な関数を構築できます。そのためには、関

297



D

E

A

B

S

C

図 3.25: 半加算器回路

数箱の出⼒を別の関数箱の⼊⼒に結線します。例えば、図 3.25に⽰す半加算器
(half-adder)は、ひとつの ORゲート、⼆つの ANDゲート、ひとつのインバー
タからなります。⼊⼒信号としては Aと Bの⼆つを取り、出⼒信号は Sと C
の⼆つです。Sは、Aと Bのどちらかひとつだけが 1のときに 1になり、Cは
Aと Bの両⽅が 1のときに 1になります。図を⾒ると、遅延があるために、出
⼒が異なる時刻に⽣成されるかもしれないということがわかります。デジタル
回路設計の設計の難しさは、多くはこの事実によって起こるものです。
ここからは、私たちの学習対象となる、デジタル論理回路をモデル化する

プログラムを作っていきます。このプログラムでは、信号を “保持”する線をモ
デル化する計算オブジェクトを構築します。関数箱は、信号同⼠の正しい関係
を成り⽴たせる⼿続きによってモデル化します。
シミュレーションの基本要素のひとつは、回路を構築する⼿続き make-

wireです。例えば、次のようにして 6つの線を構築できます。
(define a (make-wire))
(define b (make-wire))
(define c (make-wire))
(define d (make-wire))
(define e (make-wire))
(define s (make-wire))

ある関数箱を線につなげるには、そのタイプの関数箱を構築する⼿続きを呼
びます。コンストラクタ⼿続きの引数は、その箱につなげる線です。例えば、
ANDゲート、ORゲート、インバータを構築できるようになっているとする
と、図 3.25に⽰す半加算器を結線できます。
(or-gate a b d)

ok
(and-gate a b c)
ok
(inverter c e)
ok
(and-gate d e s)
ok

さらに都合のいいことに、この演算には明⽰的に名前をつけることができます。
そのためには、半加算器につなげる四本の外部の線が与えられたときにこの回
路を構築する half-adder という⼿続きを定義します。
(define (half-adder a b s c)

(let ((d (make-wire)) (e (make-wire)))
(or-gate a b d)
(and-gate a b c)
(inverter c e)
(and-gate d e s)
'ok))

この定義の利点は、より複雑な回路を作成するための構築⽤ブロックとして
half-adder⾃⾝を使うことができるということです。例えば、図 3.26は、⼆つ
の半加算器とひとつの ORゲートによって構成される全加算器 (full-adder)を
⽰したものです。26 全加算器は、次のように構築できます。
(define (full-adder a b c-in sum c-out)

(let ((s (make-wire)) (c1 (make-wire)) (c2 (make-wire)))
(half-adder b c-in s c1)
(half-adder a s sum c2)
(or-gate c1 c2 c-out)
'ok))

full-adderを⼿続きとして定義したので、今度はそれを構築⽤ブロックと
して使って、さらに複雑な回路を作ることができるようになります (例は練習
問題 3.30を参照)。

26全加算器は、⼆つの⼆進数の加算に使われる基本的な回路要素です。ここで、Aと
Bは加算する⼆つの数の対応する位置のビットで、Cin はひとつ右の位置での加算から
来るキャリー (繰り上がり)ビットです。この回路は、対応する位置の合計ビットである
SUMと、左に伝播させるキャリービットである Cout を⽣成します。

図 3.26: 全加算器回路

簡単にいうと、このシミュレータは回路を構築するための道具を提供して
くれるものだといえます。1.1節で Lispの学習に対するアプローチとして使っ
たような⼀般的な視点から⾒ると、基本関数箱は⾔語の基本要素を構成し、箱
を結線することは組み合わせの⼿段であり、⼿続きとして線のパターンを規定
することは抽象化の⼿段としての役割を果たしているといえるでしょう。

基本関数箱
基本関数箱は、ひとつの線上の信号の変化がほかの線上の信号に影響を与える
“⼒”を実装したものです。関数箱を構築するには、線に対する次のような演算
を使います。

• (get-signal ⟨wire⟩)
これは、線上の信号の現時点での値を返します。

• (set-signal! ⟨wire⟩ ⟨new value⟩)
これは、線上の信号の値を新しい値に変更します。

• (add-action! ⟨wire⟩ ⟨procedure of no arguments⟩)
これは、線上の信号の値が変わったときに、指定された⼿続きが必ず実
⾏されるようにするものです。そのような⼿続きは、線上の信号の値の
変化をほかの線に伝える⼿段となります。

これらに加え、after-delayという⼿続きも使います。これは、遅延時間と実
⾏する⼿続きを引数に取り、与えられた遅延時間後に与えられた⼿続きを実⾏
するというものです。
これらの⼿続きを使って、基本のデジタル論理関数を定義できます。イン

バータを通して⼊⼒と出⼒を接続するには、add-action!を使って、⼊⼒線上

の信号の値が変わるたびに呼ぶ⼿続きを⼊⼒線に関連づけます。この⼿続きは
⼊⼒信号の logical-not(論理否定)を計算し、そして inverter-delay1単位
後に、出⼒信号にその新しい値をセットします。
(define (inverter input output)

(define (invert-input)
(let ((new-value (logical-not (get-signal input))))

(after-delay inverter-delay
(lambda () (set-signal! output new-value)))))

(add-action! input invert-input) 'ok)
(define (logical-not s)

(cond ((= s 0) 1)
((= s 1) 0)
(else (error "Invalid signal" s))))

ANDゲートはもうちょっと複雑です。アクション⼿続きは、⼊⼒のどちらか
ひとつが変化したときに動くようにする必要があります。アクション⼿続きは、
⼊⼒線上の信号の値の logical-and(論理積)を (logical-notと似たような⼿
続きを使って)求め、and-gate-delay1単位後に出⼒線上で新しい値への変化
が起こるように設定します。
(define (and-gate a1 a2 output)

(define (and-action-procedure)
(let ((new-value

(logical-and (get-signal a1) (get-signal a2))))
(after-delay
and-gate-delay
(lambda () (set-signal! output new-value)))))

(add-action! a1 and-action-procedure)
(add-action! a2 and-action-procedure)
'ok)

練習問題 3.28: ORゲートを基本関数箱として定義せよ。or-gate
のコンストラクタは and-gateと似たようなものにすること。
練習問題 3.29: ORゲートを構築する別の⽅法として、ANDゲー
トとインバータからなる複合デジタル論理デバイスとして組み⽴
てるというものがある。これを実現する⼿続き or-gateを定義せ

図 3.27: nbitの数を扱う繰り上がり伝播加算器

よ。この ORゲートの遅延時間は、and-gate-delayと inverter-
delayによって表すと、どのようになるだろうか。
練習問題 3.30: 図 3.27は、n個の全加算器をつなげた繰り上がり
伝播加算器 (ripple-carry adder)を⽰している。これは、⼆つの n

ビット⼆進数を⾜すための最も単純な形の並列加算器である。A1,
A2, A3, . . ., An と B1, B2, B3, . . ., Bn という⼊⼒は、⾜し合わせ
る⼆つの⼆進数 (各 Ak, Bk は 0または 1)である。この回路は、和
のうちの nビットの S1, S2, S3, . . ., Sn と、⾜し算による繰り上
がりの C を⽣成する。この回路を⽣成する⼿続き ripple-carry-
adder を書け。この⼿続きは、n 本の線のリスト三つ— Ak, Bk,
Sk—と、もうひとつの線 C を引数として取るようにする。繰り上
がり伝播加算器の⼤きな⽋点は、キャリー信号が伝播するのを待
つ必要があるということだ。nビットの繰り上がり伝播加算器か
ら完全な出⼒を得るのにかかる遅延時間は、ANDゲート・ ORゲ
ート・インバータの遅延時間によって表現すると、どのようにな
るだろうか。

回路の表現
このシミュレーションでは、線は⼆つの局所状態変数を持つ計算オブジェクト
になります。信号の値 signal-value(初期値は 0)と、信号の値が変わるときに
実⾏するアクション⼿続き action-proceduresの集合です。メッセージパッ
シングスタイルを使い、局所⼿続きの集合に適切な局所演算を選ぶ dispatch
⼿続きを加えたものとして線を実装します。これは、3.1.1節で単純な銀⾏⼝座
オブジェクトを実装したときと同じです。

(define (make-wire)
(let ((signal-value 0) (action-procedures '()))

(define (set-my-signal! new-value)
(if (not (= signal-value new-value))

(begin (set! signal-value new-value)
(call-each action-procedures))

'done))
(define (accept-action-procedure! proc)

(set! action-procedures
(cons proc action-procedures))

(proc))
(define (dispatch m)

(cond ((eq? m 'get-signal) signal-value)
((eq? m 'set-signal!) set-my-signal!)
((eq? m 'add-action!) accept-action-procedure!)
(else (error "Unknown operation: WIRE" m))))

dispatch))

局所⼿続き set-my-signal!は、新しい信号値によって線上の信号が変わるか
どうかをテストします。変わるのであれば、以下の call-each⼿続きを使って
各アクション⼿続きを実⾏します。call-eachは、引数なし⼿続きのリストの
各項⽬を呼び出すというものです。
(define (call-each procedures)

(if (null? procedures)
'done
(begin ((car procedures))

(call-each (cdr procedures)))))

局所⼿続き accept-action-procedure! は、与えられた⼿続きを実⾏する⼿
続きのリストに追加し、それから新しい⼿続きを⼀回実⾏します (練習問題
3.31参照)。
局所⼿続き dispatchが上で規定した通りに設定されていれば、以下のいく

つかの⼿続きを提供することで線に対する局所演算にアクセスできるようにな
ります。27

27これらの⼿続きは、通常の⼿続き的構⽂によってオブジェクトの局所⼿続きにアク
セスできるようにするシンタックスシュガーにすぎません。“⼿続き”と “データ”の役

(define (get-signal wire) (wire 'get-signal))
(define (set-signal! wire new-value)

((wire 'set-signal!) new-value))
(define (add-action! wire action-procedure)

((wire 'add-action!) action-procedure))

線は、時とともに変化する信号を持ち、接続される装置が増えていくこともあ
るので、可変オブジェクトの典型といえます。ここでは、代⼊によって変化す
る局所状態変数を持つ⼿続きとして線をモデル化しています。新しい線が作ら
れると、状態変数の新しいセットが (make-wire中の let式によって) 割り当
てられ、それらの新しい状態変数を含む環境を持つ新しい dispatch⼿続きが
構築され、それが返されることになります。
線は、それに接続されたいろいろなデバイスの間で共有されます。そのた

め、ひとつのデバイスとのやりとりによって起こった変化が、その線につなが
ったほかのすべてのデバイスに影響することになります。線は、接続時に提供
されるアクション⼿続きを呼ぶことによって、変化を近隣に伝えます。

予定表
後は、after-delayさえあればシミュレータが完成します。ここでの考え⽅は、
するべきことのスケジュールを持っておく予定表 (agenda)と呼ばれるデータ
構造を保持するということです。予定表に対しては、以下の演算が定義され
ます。

• (make-agenda)は、新しい空の予定表を返します。
• (empty-agenda? ⟨agenda⟩)は、指定した予定表が空のときに真になり
ます。

• (first-agenda-item ⟨agenda⟩)は、予定表の最初の項⽬を返します。
• (remove-first-agenda-item! ⟨agenda⟩) は、最初の項⽬を削除するこ
とによって、予定表を変更します。

割がこんなに簡単なやり⽅で交換できるということは特筆すべきことです。例えば、も
し (wire 'get-signal)と書くようにすると、wireは get-signalというメッセージを
⼊⼒として呼び出す⼿続きとして考えるということになります。(get-signal wire)と
書くようにすると、wireは⼿続き get-signalの⼊⼒となるデータオブジェクトだと考
えやすくなります。この問題の実際のところは、⼿続きをオブジェクトとして扱う⾔語
においては “⼿続き”と “データ”の間に根本的な違いはなく、好きなスタイルでプログ
ラミングできるようにシンタックスシュガーを選ぶことができるということです。

• (add-to-agenda! ⟨time⟩ ⟨action⟩ ⟨agenda⟩)は、指定された時間に実
⾏するアクション⼿続きを追加することによって、予定表を変更します。

• (current-time ⟨agenda⟩)は、現在のシミュレーション時間を返します。
ここで使う特定の予定表は、the-agendaとして⽰します。⼿続き after-delay
は、the-agendaに新しい要素を追加します。
(define (after-delay delay action)

(add-to-agenda! (+ delay (current-time the-agenda))
action
the-agenda))

シミュレーションを動かすのは、⼿続き propagate(伝播)です。この⼿続きは
the-agendaを操作し、この予定表に載っているそれぞれの⼿続きを順番に実
⾏します。⼀般的に、シミュレーションが進むにつれ、新しい項⽬が予定表に
追加されていき、propagateは予定表に項⽬が存在する限りずっとシミュレー
ションを続けます。
(define (propagate)

(if (empty-agenda? the-agenda)
'done
(let ((first-item (first-agenda-item the-agenda)))

(first-item)
(remove-first-agenda-item! the-agenda)
(propagate))))

シミュレーション例
次の⼿続きは、線に “probe”(プローブ)を置くというもので、シミュレータの
動作を表⽰します。プローブは、信号値が変わるたびに新しい信号値・現在時
刻・線の識別名を表⽰するよう線に伝えます。
(define (probe name wire)

(add-action! wire
(lambda ()

(newline)
(display name) (display " ")
(display (current-time the-agenda))

(display " New-value = ")
(display (get-signal wire)))))

まず、予定表を初期化し、基本関数箱の遅延時間を規定します。
(define the-agenda (make-agenda))
(define inverter-delay 2)
(define and-gate-delay 3)
(define or-gate-delay 5)

次に、四つの配線を定義し、そのうち⼆つにはプローブを設置します。
(define input-1 (make-wire))
(define input-2 (make-wire))
(define sum (make-wire))
(define carry (make-wire))

(probe 'sum sum)
sum 0 New-value = 0
(probe 'carry carry)
carry 0 New-value = 0

それから、線を (図 3.25のように)半加算器回路に接続し、input-1の信号を 1
に設定し、シミュレーションを実⾏します。
(half-adder input-1 input-2 sum carry)
ok

(set-signal! input-1 1)
done

(propagate)
sum 8 New-value = 1
done

sumの信号は、時刻 8で 1に変わります。シミュレーション開始から 8単位時
間がたったことになります。ここで、input-2の信号を 1に設定し、値が伝播
するようにします。
(set-signal! input-2 1)
done

(propagate)
carry 11 New-value = 1
sum 16 New-value = 0
done

carryは時刻 11で 1に変わり、sumは時刻 16で 0に変わります。
練習問題 3.31: make-wireの中で定義される内部⼿続き accept-
action-procedure!は、新しいアクション⼿続きが線に追加され
るときに、その⼿続きがすぐに実⾏されるということを規定して
いる。この初期化がなぜ必要なのか説明せよ。具体的には、上の段
落の半加算器の例をトレースし、accept-action-procedure!を以
下のように定義した場合にシステムの応答がどのように変わるか
述べよ。
(define (accept-action-procedure! proc)

(set! action-procedures
(cons proc action-procedures)))

予定表の実装
最後に、予定表のデータ構造の詳細を記述します。このデータ構造は、将来実
⾏するようスケジュールされた⼿続きを持っておくものです。
予定表は、複数の時間区分 (time segment)によって構成されます。それぞ

れの時間区分は、数値 (時刻)と、その時間区分の間に実⾏するようスケジュー
ルされた⼿続きを持っておくキュー (練習問題 3.32参照)からなるペアです。
(define (make-time-segment time queue)

(cons time queue))
(define (segment-time s) (car s))
(define (segment-queue s) (cdr s))

時間区分キューは、3.3.2節で説明したキュー演算によって操作します。
予定表⾃⾝は、時間区分の⼀次元テーブルです。3.3.3節で説明したテーブ

ルと違うのは、時間区分は時間の前後順に並んでいるという点です。それに加
えて、予定表の先頭には 現在時刻 (current time)(最後に処理したアクションの

時刻)を保存します。新しく構築したばかりの予定表には時間区分はなく、現
在時刻としては 0を持っています。28

(define (make-agenda) (list 0))
(define (current-time agenda) (car agenda))
(define (set-current-time! agenda time)

(set-car! agenda time))
(define (segments agenda) (cdr agenda))
(define (set-segments! agenda segments)

(set-cdr! agenda segments))
(define (first-segment agenda) (car (segments agenda)))
(define (rest-segments agenda) (cdr (segments agenda)))

予定表に時間区分がなければ、その予定表は空だということです。
(define (empty-agenda? agenda)

(null? (segments agenda)))

予定表にアクションを追加するには、まず予定表が空かどうかチェックします。
空であれば、アクションの時間区分を作り、それを予定表に組み込みます。空
でなければ、それぞれの時間区分の時刻を調べながら予定表を⾛査します。指
定時刻に時間区分が⾒つかれば、それに関連づけられたキューにアクションを
追加します。指定時刻以降の時刻にたどり着いたら、その時刻のすぐ前に、新
しい時間区分を挿⼊します。予定表の最後にたどり着いたら、末尾に新しい時
間区分を作る必要があります。
(define (add-to-agenda! time action agenda)

(define (belongs-before? segments)
(or (null? segments)

(< time (segment-time (car segments)))))
(define (make-new-time-segment time action)

(let ((q (make-queue)))
(insert-queue! q action)
(make-time-segment time q)))

(define (add-to-segments! segments)
(if (= (segment-time (car segments)) time)

28予定表は3.3.3節のような頭つきリストですが、このリストには時刻というヘッダが
あるので、追加のダミーヘッダ (テーブルで使った *table*記号のようなもの) の必要は
ありません。

(insert-queue! (segment-queue (car segments))
action)

(let ((rest (cdr segments)))
(if (belongs-before? rest)

(set-cdr!
segments
(cons (make-new-time-segment time action)

(cdr segments)))
(add-to-segments! rest)))))

(let ((segments (segments agenda)))
(if (belongs-before? segments)

(set-segments!
agenda
(cons (make-new-time-segment time action)

segments))
(add-to-segments! segments))))

予定表の先頭の項⽬を削除する⼿続きは、先頭の時間区分にあるキューの先頭
の項⽬を削除します。削除して時間区分が空になるなら、それを時間区分のリ
ストから削除します。29

(define (remove-first-agenda-item! agenda)
(let ((q (segment-queue (first-segment agenda))))

(delete-queue! q)
(if (empty-queue? q)

(set-segments! agenda (rest-segments agenda)))))

予定表の先頭の項⽬は、先頭の時間区分のキューの先頭にあります。項⽬を取
り出すたびに、現在時刻の更新も⾏います。30

(define (first-agenda-item agenda)
(if (empty-agenda? agenda)

29この⼿続きの if式には ⟨alternative⟩式がありません。このような “⽚腕 if⽂”は、
⼆つの式から選ぶというのではなく、何かをするかどうかを決めるという場合に使いま
す。if式は、述語が偽であり、かつ ⟨alternative⟩が存在しない場合、不定の値を返しま
す。

30こうすると、現在時刻は常に直近に処理したアクションの時刻になります。この時刻
を予定表の先頭に格納することで、関連づけられた時間区分が削除されていてもその時
刻が確実にわかるようになっています。

(error "Agenda is empty: FIRST-AGENDA-ITEM")
(let ((first-seg (first-segment agenda)))

(set-current-time! agenda
(segment-time first-seg))

(front-queue (segment-queue first-seg)))))

練習問題 3.32: 予定表の各時間区分で実⾏するべき⼿続きはキュ
ーに保存される。そのため、各時間区分の⼿続きは、予定表に追加
した順 (最初に⼊れたものが最初に取り出される)で呼び出される。
どうしてこの順番でなければならないのか説明せよ。具体的には、
ひとつの時間区分の中で⼊⼒が 0, 1から 1, 0に変わる ANDゲー
トのふるまいをトレースし、時間区分の⼿続きを通常のリストと
して先頭だけで追加と削除を⾏う (最後に⼊れたものが最初に取
り出される)ようにするとふるまいがどのように変わるか述べよ。

3.3.5 制約伝播
コンピュータプログラムは伝統的に⼀⽅向の計算という構造になっていて、あ
らかじめ決められた引数に演算を⾏い、望む出⼒を⽣成します。⼀⽅で、いく
つかの量の間の関係を使ってシステムをモデル化するということもよくありま
す。例えば、機械構造の数学的モデルには、⾦属棒のたわみ dと棒に加わる⼒
F、棒の⻑さ L、断⾯積 A、弾性係数 E との間に成り⽴つ、次の⽅程式によっ
て表される関係についての情報が含まれるかもしれません。

dAE = FL.

このような⽅程式は⼀⽅向ではありません。量のうち任意の四つがあれば、そ
れを使って五つ⽬の量を計算できます。しかし、この⽅程式を伝統的なコンピ
ュータ⾔語に翻訳しようとすると、どの量を残り四つの量を使って計算するの
か選ばなければなりません。そのため、断⾯積 Aとたわみ dの計算は同じ⽅程
式から導かれるものなのに、Aを計算する⼿続きは dを求めるのには使えない
ということになります。31

31制約伝播は、Ivan Sutherland (1963)による、信じられないほどの先進性を持っ
た sketchpad システムで初登場したものです。Xerox パロアルト研究センターでは、
Smalltalkベースの美しい制約伝播システムが Alan Borning (1977)によって開発されま
した。Sussman, Stallman, Steeleは、制約伝播を電⼦回路解析に応⽤しました (Sussman
and Stallman 1975; Sussman and Steele 1980)。TK!Solver(Konopasek and Jayaraman

この節では、関係そのものによってプログラミングできるような⾔語の
設計についての概略を述べます。この⾔語の基本要素は基本制約 (primitive
constraint)です。これは、量と量の間に成り⽴つ関係を記述したものです。例
えば、(adder a b c)は a, b, cという量の間に a+ b = cという⽅程式の関係
が成り⽴っていなければならないということを規定し、(multiplier x y z)
は、xy = zという制約を表し、(constant 3.14 x)は xの値が 3.14でなけれ
ばならないということを表しています。
この⾔語では、基本制約を組み合わせてより複雑な関係を表す⽅法を提供

します。制約を組み合わせるには、コネクタ (connector)を使って制約を接続
し、制約ネットワーク (constraint network)を構築します。コネクタは、最低ひ
とつの制約に含まれる値を “保持する”オブジェクトです。例を挙げると、華
⽒温度と摂⽒温度には次のような関係があるということを私たちは知ってい
ます。

9C = 5(F − 32).

このような制約は、加算器・乗算器・不変制約という基本制約からなるネット
ワーク (図 3.28)として考えることができます。図の中で、左には m1, m2, p
というラベルのついた三つの端⼦を持つ乗算箱があります。これらの端⼦は、
次のようにして乗算器をネットワークのほかの部分と接続します。まず、m1
端⼦は摂⽒の気温を保持するコネクタ C につなげられます。m2端⼦は wとい
うコネクタにつなげられます。このコネクタは 9を保持する定数箱にもつなが
っています。p端⼦は乗算箱によって m1と m2の積になるよう制約されてい
て、別の乗算箱の p端⼦につながっています。この乗算箱では、m2は整数 5
に、m1は加算の項のうちのひとつに接続されています。
このようなネットワークによる計算の進⾏は以下のようになります。コネ

クタに (ユーザーまたは接続された制約箱によって)値が与えられると、コネ
クタは関連づけられた制約すべて (そのコネクタを起動した制約を除き)を起
動し、値が与えられたことを知らせます。起動されたそれぞれの制約箱は、そ
れに接続されているコネクタを調査して、コネクタの値を決めるのに⼗分な情
報があるかどうかを⾒ます。もし値が決められるなら、制約箱はそのコネクタ
に値を設定し、するとそのコネクタは関連づけられた制約すべてを起動し . . .

という連鎖が続きます。例えば、摂⽒と華⽒の変換の場合、制約箱によって w,
x, yはそれぞれ 9, 5, 32にすぐに設定されます。それらのコネクタは乗算器と
加算器を起動し、それらは計算を進⾏させるために⼗分な情報がないと判断し
ます。もしユーザー (またはネットワークのどこか別の場所)が C にある値 (例
1984)は、制約に基づく⼤規模モデリング環境です。

図 3.28: 制約ネットワークとして表現された関係
9C = 5(F − 32)

えば 25)を設定すると、左端の乗算器が起動され、uに 25 · 9 = 225を設定しま
す。すると uは⼆つ⽬の乗算器を起動し、それが v に 45を設定します。そし
て vが加算器を起動し、加算器は f を 77に設定します。

制約システムを使う
上で概要を述べた温度計算を制約システムを使って実⾏するには、まずコンス
トラクタ make-connectorを呼んで C, Fという⼆つのコネクタを作り、それら
を適切なネットワークに接続します。
(define C (make-connector))
(define F (make-connector))
(celsius-fahrenheit-converter C F)
ok

ネットワークを作る⼿続きは、次のように定義します。
(define (celsius-fahrenheit-converter c f)

(let ((u (make-connector))
(v (make-connector))
(w (make-connector))
(x (make-connector))
(y (make-connector)))

(multiplier c w u)
(multiplier v x u)
(adder v y f)

(constant 9 w)
(constant 5 x)
(constant 32 y)
'ok))

この⼿続きは内部コネクタ u, v, w, x, y を作成し、基本制約のコンストラク
タ adder, multiplier, constantを使って 図 3.28に⽰したように接続します。
3.3.4節のデジタル回路シミュレータの場合と同じように、基本要素の組み合
わせに⼿続きを使うことによって、この⾔語は⾃動的に複合オブジェクトを抽
象化する⼿段を持つことになります。
ネットワークの動作を⾒るために、3.3.4節で配線の監視に使ったのと似た

ような probe⼿続きを使って、コネクタ C, Fにプローブを置きます。コネクタ
にプローブを置くと、コネクタに値が与えられるたびにメッセージを表⽰する
ようになります。
(probe "Celsius temp" C)
(probe "Fahrenheit temp" F)

それから、Cの値を 25に設定します。(set-value!への三つ⽬の引数は、この
指⽰が userによるものであることを Cに伝えています)。
(set-value! C 25 'user)
Probe: Celsius temp = 25
Probe: Fahrenheit temp = 77
done

Cのプローブが起動され、値を報告します。Cはまた、値を上記のネットワー
クに伝播させます。それによって Fに 77が設定され、それが Fのプローブに
よって報告されます。
今度は、Fに新しい値、例えば 212を設定してみます。

(set-value! F 212 'user)
Error! Contradiction (77 212)

コネクタは⽭盾を検知したというメッセージを出します。値は 77なのに、誰
かがそれを 212に設定しようとしているということです。新しい値によってネ
ットワークを再利⽤するには、Cに古い値を忘れるよう指⽰します。
(forget-value! C 'user)
Probe: Celsius temp = ?

Probe: Fahrenheit temp = ?
done

Cは、最初に値を設定した userがその値を取り下げたことを検知し、Cはその
値を捨てることに同意します。プローブの結果はそのことを⽰しています。そ
して、Cはネットワークのほかの部分にその事実を伝えます。この情報は最終
的に Fにまで伝播します。Fは、⾃分の値が 77だと信じ続ける根拠がなくなっ
たことに気づきます。このようにして、Fもまた⾃分の値を捨てることになり
ます。これもプローブによって表⽰されています。
これで F は値を持たなくなったので、F に 212 を設定できるようになり

ます。
(set-value! F 212 'user)
Probe: Fahrenheit temp = 212
Probe: Celsius temp = 100
done

この新しい値がネットワークを通して伝播されると、Cの値が 100に設定され、
それが Cのプローブに記録されます。Fから Cを計算するのにも、Cから Fを
計算するのにも、完全に同じネットワークが使われていることに注意してくだ
さい。この計算の無⽅向性は、制約ベースシステムの際⽴った特徴です。

制約システムの実装
制約システムは、3.3.4節のデジタル回路シミュレータとよく似たやり⽅で、局
所状態を持つ⼿続きオブジェクトによって実装します。制約システムの基本オ
ブジェクトはもう少し複雑なものになりますが、予定表や論理遅延について考
える必要がない分、システム全体としてはこちらのほうが単純です。
コネクタに対する基本演算は次の通りです。
• (has-value? <connector>)は、コネクタが値を持っているかどうかを
返します。

• (get-value <connector>)は、コネクタの現在の値を返します。
• (set-value! <connector> <new-value> <informant>) は、informant
がコネクタに対して新しい値を設定するよう要求しているということを
⽰しています。

• (forget-value! <connector> <retractor>)は、retractor が現在の値
を忘れるよう要求しているということをコネクタに対して伝えます。

• (connect <connector> <new-constraint>)は、新しい制約に参加する
ようコネクタに伝えます。

コネクタは、⼿続き inform-about-valueと inform-about-no-valueによ
って制約と通信します。inform-about-valueは与えられた制約にコネクタが
値を持ったことを伝え、inform-about-no-valueは制約にコネクタが値を失っ
たことを伝えます。

adderは、a1, a2という加数コネクタと sumコネクタとの間に加算器制約
を構築するコンストラクタです。加算器は、局所状態 (下記の me⼿続き)を持
つ⼿続きとして実装します。
(define (adder a1 a2 sum)

(define (process-new-value)
(cond ((and (has-value? a1) (has-value? a2))

(set-value! sum
(+ (get-value a1) (get-value a2))
me))

((and (has-value? a1) (has-value? sum))
(set-value! a2

(- (get-value sum) (get-value a1))
me))

((and (has-value? a2) (has-value? sum))
(set-value! a1

(- (get-value sum) (get-value a2))
me))))

(define (process-forget-value)
(forget-value! sum me)
(forget-value! a1 me)
(forget-value! a2 me)
(process-new-value))

(define (me request)
(cond ((eq? request 'I-have-a-value) (process-new-value))

((eq? request 'I-lost-my-value) (process-forget-value))
(else (error "Unknown request: ADDER" request))))

(connect a1 me)
(connect a2 me)
(connect sum me)

me)

adderは、新しい加算器を指定されたコネクタに接続し、その加算器を値とし
て返します。加算器を表現する⼿続き meは、局所⼿続きを起動するディスパ
ッチの役割を果たしています。
(define (inform-about-value constraint)

(constraint 'I-have-a-value))
(define (inform-about-no-value constraint)

(constraint 'I-lost-my-value))

加算器の局所⼿続き process-new-valueは、コネクタのうちひとつが値を持
ったことが加算器に伝えられたときに呼び出されます。加算器は、まず a1と
a2の両⽅が値を持っているかチェックします。持っているなら、⼆つの加数の
和を値として設定するよう sumに伝えます。set-value!の informant引数は、
加算器オブジェクト⾃⾝である meになります。a1と a2の両⽅が値を持ってい
るわけではない場合、a1と sumが値を持っているかもしれないので、加算器は
それをチェックします。もしそうなら、a2にはその⼆つの差を設定します。最
後に、a2と sumが値を持っているなら、加算器には a1の値を設定するのに⼗
分な情報があることになります。コネクタのひとつが値を失ったことが加算器
に伝えられると、加算器はすべてのコネクタに値を捨てるよう要求します (実
際に捨てられるのはこの加算器によって設定された値だけです)。それから、加
算器は process-new-valueを実⾏します。この最後のステップの理由は、ひと
つ以上のコネクタがまだ値を持っているかもしれず (つまり、元々その加算器
によって設定されたのではない値を持っていたということ)、それらの値は加
算器を通して伝播させ直す必要があるかもしれないからです。
乗算器は加算器に⾮常によく似ています。乗算器は、因数の⽚⽅が 0なら、

もう⽚⽅の値がわからなくても productを 0に設定します。
(define (multiplier m1 m2 product)

(define (process-new-value)
(cond ((or (and (has-value? m1) (= (get-value m1) 0))

(and (has-value? m2) (= (get-value m2) 0)))
(set-value! product 0 me))

((and (has-value? m1) (has-value? m2))
(set-value! product

(* (get-value m1) (get-value m2))
me))

((and (has-value? product) (has-value? m1))
(set-value! m2

(/ (get-value product)
(get-value m1))

me))
((and (has-value? product) (has-value? m2))
(set-value! m1

(/ (get-value product)
(get-value m2))

me))))
(define (process-forget-value)

(forget-value! product me)
(forget-value! m1 me)
(forget-value! m2 me)
(process-new-value))

(define (me request)
(cond ((eq? request 'I-have-a-value) (process-new-value))

((eq? request 'I-lost-my-value) (process-forget-value))
(else (error "Unknown request: MULTIPLIER"

request))))
(connect m1 me)
(connect m2 me)
(connect product me)
me)

constantコンストラクタは、単純に指定されたコネクタの値を設定します。I-
have-a-valueや I-lost-my-valueのメッセージが定数箱に送られるとエラー
を発⽣させます。
(define (constant value connector)

(define (me request)
(error "Unknown request: CONSTANT" request))

(connect connector me)
(set-value! connector value me)
me)

最後に、プローブは指定されたコネクタに値が設定されたり解除されたりした
ことについてのメッセージを表⽰します。

(define (probe name connector)
(define (print-probe value)

(newline) (display "Probe: ") (display name)
(display " = ") (display value))

(define (process-new-value)
(print-probe (get-value connector)))

(define (process-forget-value) (print-probe "?"))
(define (me request)

(cond ((eq? request 'I-have-a-value) (process-new-value))
((eq? request 'I-lost-my-value) (process-forget-value))
(else (error "Unknown request: PROBE" request))))

(connect connector me)
me)

コネクタの表現
コネクタは⼿続きオブジェクトとして表現します。局所状態変数として、コネ
クタの現在の値 value、コネクタの値を設定したオブジェクト informant、コ
ネクタが参加する制約のリスト constraintsを持ちます。
(define (make-connector)

(let ((value false) (informant false) (constraints '()))
(define (set-my-value newval setter)

(cond ((not (has-value? me))
(set! value newval)
(set! informant setter)
(for-each-except setter

inform-about-value
constraints))

((not (= value newval))
(error "Contradiction" (list value newval)))

(else 'ignored)))
(define (forget-my-value retractor)

(if (eq? retractor informant)
(begin (set! informant false)

(for-each-except retractor

inform-about-no-value
constraints))

'ignored))
(define (connect new-constraint)

(if (not (memq new-constraint constraints))
(set! constraints

(cons new-constraint constraints)))
(if (has-value? me)

(inform-about-value new-constraint))
'done)

(define (me request)
(cond ((eq? request 'has-value?)

(if informant true false))
((eq? request 'value) value)
((eq? request 'set-value!) set-my-value)
((eq? request 'forget) forget-my-value)
((eq? request 'connect) connect)
(else (error "Unknown operation: CONNECTOR"

request))))
me))

コネクタの局所⼿続き set-my-valueは、コネクタの値を設定するよう要求さ
れたときに呼ばれます。コネクタが現時点で値を持っていなければ、その値を
設定し、値の設定を要求してきた制約を informantとして覚えておきます。32

それからコネクタは、値の設定を要求してきた制約を除き、参加している制約
すべてに通知を⾏います。これは、次のイテレータによって実現します。この
イテレータは、与えられたひとつを除いて、リストのすべての項⽬に対して指
定された⼿続きを適⽤します。
(define (for-each-except exception procedure list)

(define (loop items)
(cond ((null? items) 'done)

((eq? (car items) exception) (loop (cdr items)))
(else (procedure (car items))

(loop (cdr items)))))
(loop list))

32setterは制約とは限りません。温度の例では、userを setterとして使いました。

コネクタが値を忘れるよう依頼されると、局所⼿続き forget-my-valueを実⾏
します。この⼿続きは、その要求が元々その値を設定したオブジェクトと同じ
オブジェクトから来たものかどうかを確認します。そうであれば、コネクタは
関連づけられた制約に値を失ったことを伝えます。
局所⼿続き connectは、もし制約が制約リストにすでにあるのでなければ、

制約リストに新しい制約を加えます。それから、コネクタが値を持っているな
ら、新しい制約にその事実を伝えます。
コネクタの⼿続き meは、ほかの内部⼿続きへのディスパッチを⾏う役割を

果たし、またコネクタのオブジェクトとしての表現にもなっています。以下の
⼿続きは、ディスパッチ⽤の構⽂インターフェイスを提供します。
(define (has-value? connector)

(connector 'has-value?))
(define (get-value connector)

(connector 'value))
(define (set-value! connector new-value informant)

((connector 'set-value!) new-value informant))
(define (forget-value! connector retractor)

((connector 'forget) retractor))
(define (connect connector new-constraint)

((connector 'connect) new-constraint))

練習問題 3.33: 三つのコネクタ a, b, cを⼊⼒として取り、cの値
が aと bの値の平均となるような制約を作る⼿続き averagerを、
基本の乗算器・加算器・定数制約を使って定義せよ。
練習問題 3.34: Louis Reasonerは、squarerという制約を作りたい
と思っている。squarerは⼆つの端⼦を持ち、⼆つ⽬の端⼦のコネ
クタ bは常に⼀つ⽬の端⼦の値 aの⼆乗になる。彼は、ひとつの
乗算器からなる次の単純な装置を提案した。
(define (squarer a b)

(multiplier a a b))

このアイデアには重⼤な⽋陥がある。説明せよ。
練習問題 3.35: Ben Bitdiddleは Louisに、練習問題 3.34の問題を
避ける⽅法のひとつは、squarerを新しい基本制約として定義する
ことだと伝えた。Benが書いた⼿続き概略の⽋けている部分を埋
め、そのような制約を実装せよ。

(define (squarer a b)
(define (process-new-value)

(if (has-value? b)
(if (< (get-value b) 0)

(error "square less than 0: SQUARER"
(get-value b))

⟨alternative1⟩)
⟨alternative2⟩))

(define (process-forget-value) ⟨body1⟩)
(define (me request) ⟨body2⟩)
⟨rest of definition⟩
me)

練習問題 3.36: グローバル環境で、次のような式の列を評価する
とする。
(define a (make-connector))
(define b (make-connector))
(set-value! a 10 'user)

set-value!の評価中のある時点で、コネクタの局所⼿続きの中の
以下の式が評価されることになる。
(for-each-except

setter inform-about-value constraints)

上記の式が評価される環境を表す環境図を描け。
練習問題 3.37: celsius-fahrenheit-converter⼿続きは、以下の
ような式指向スタイルによる定義に⽐べると⾯倒だ。
(define (celsius-fahrenheit-converter x)

(c+ (c* (c/ (cv 9) (cv 5))
x)

(cv 32)))
(define C (make-connector))
(define F (celsius-fahrenheit-converter C))

ここで、c+, c*等は “制約”バージョンの数値演算である。例えば、
c+は⼆つのコネクタを引数として取り、加算器制約によってそれ
らと関連づけられたコネクタを返す。

(define (c+ x y)
(let ((z (make-connector)))

(adder x y z)
z))

似たような⼿続き c-, c*, c/, cv(定数)を定義し、上記の変換器の
例のように複合制約を定義できるようにせよ。33

3.4 並⾏性: 期限厳守
ここまで、モデル化のための道具として、局所状態を持つ計算オブジェクトと
いうものが持つ⼒について⾒てきました。しかし、3.1.3節で警告したように、

33式指向形式を使うと計算の中間の式に名前をつける必要がなくなるので便利です。元
々の制約⾔語による定式化は⾯倒でした。これは、多くの⾔語で複合データを扱うのが
⾯倒なのと同じです。例えば、変数がベクトルを表現しているとして、積 (a+ b) · (c+ d)
を求めたいとすると、指定されたベクトル引数に値をセットするが⼿続き⾃⾝はベクト
ルを値として返さないような⼿続きを使って、“命令型スタイル”で作業をすることもで
きます。
(v-sum a b temp1)
(v-sum c d temp2)
(v-prod temp1 temp2 answer)

別の⽅法として、ベクトルを値として返す⼿続きを使うことによって、式によって扱
うことも可能です。こうすると、明⽰的に temp1と temp2を使う必要がなくなります。
(define answer (v-prod (v-sum a b) (v-sum c d)))

Lispでは、⼿続きの値として複合オブジェクトを返すことができるので、ここまで命
令型スタイルで作ってきた制約⾔語を、この練習問題のように式指向のスタイルに変形
するということができます。複合データを扱う⼿段に乏しい Algol, Basic, Pascal(明⽰
的に Pascalのポインタ変数を使わない限り) では、複合オブジェクトを操作しようと思
うと、命令型スタイルに縛られることになります。式指向形式の利点を考えると、この
節でやったように命令型スタイルでシステムを実装する意味はあるのかと思う⼈もいる
かもしれません。命令型スタイルで実装した理由のひとつは、式指向でない制約⾔語は
コネクタオブジェクトだけでなく制約オブジェクトに対するハンドル (例えば、adder⼿
続きの値)を提供するということです。コネクタ演算を通して間接的に制約を扱うので
はなく、制約と直接通信する新しい演算によってシステムを拡張したいという場合には、
このことが役に⽴ちます。命令型の実装によって式指向スタイルを実装するのは簡単な
のですが、その逆はとても難しいのです。

この⼒は代償を伴うものです。その代償とは、参照透明性を失い、結果として
同⼀性と変化ということについて厄介な問題を引き起こしたこと、評価の置換
モデルを放棄し、より複雑な環境モデルを採⽤せざるを得なくなったことです。
状態、等価性、変化の複雑性の下に潜んでいる中⼼的課題は、代⼊を導⼊

することによって計算モデルの中に時間 (time)という概念を含めなければな
らなくなったということです。代⼊を導⼊するまでは、値を持つ式はいつでも
同じ値を持つという意味で、プログラムには時間という概念がありませんでし
た。それに対して、3.1.1節の最初で導⼊した、銀⾏⼝座から引き出しを⾏い差
引残⾼を返すということのモデル化の例を思い出してみましょう。
(withdraw 25)
75
(withdraw 25)
50

ここでは、同じ式を連続して評価して、違う値が返されています。このふるま
いは、代⼊⽂ (この場合、変数 balanceへの代⼊)を実⾏することが、値が変化
する 瞬間 (moments in time)を⽰していることによります。ある式を評価した
結果は、式そのもの以外に、その評価が起こったのがこれらの瞬間の前なのか、
それとも後なのかということによっても変わることになります。局所状態を持
つ計算オブジェクトによってモデルを構築すると、プログラミングの本質的な
概念として、時間という問題に取り組む必要が出てきます。
物理世界の認知に合うように、計算モデルの構造化をさらに進めることも

できます。現実世界のオブジェクトは、ひとつずつ順番に変化するということ
はありません。オブジェクトは、並⾏ (concurrently)に、つまり同時に動いて
いるように認識されています。そのため、システムをモデル化する際に、並⾏
に動く計算プロセスの集合という形で⾏うことが⾃然になることはよくありま
す。別々の局所状態を持ったオブジェクトを使ってモデルを作り上げることに
よってプログラムをモジュール式にできるのと同じように、独⽴して並⾏に実
⾏が進むいくつかの部品に計算モデルを分割するのが向いているということが
よくあります。プログラムは逐次型コンピュータで実⾏することになるとして
も、あたかも並⾏に実⾏されるかのようにプログラムを書くと、プログラマは
本質的でない時間の制約を避けるようになり、そのためプログラムのモジュー
ル性が⾼まります。
プログラムのモジュール性を⾼めるほかに、並⾏計算は逐次計算に対して

スピード的な利点もあります。逐次計算は⼀度にひとつの演算しか実⾏できな

323



いので、タスクの実⾏時間は実⾏する演算の数に⽐例します。34 しかし、もし
問題を⽐較的独⽴した、時々しか通信を⾏わないいくつかの部品に分割できる
なら、それぞれの部品を別々のプロセッサに割り当てられる可能性があります。
そうすると、使えるプロセッサの数に⽐例して速度を上げることができるよう
になります。
残念ながら、並⾏性という概念を持ち込むと、代⼊によって⽣じた複雑性

がさらに問題のあるものになります。並⾏実⾏という現象は、現実世界が並列
に動いているからという理由であれ、コンピュータが並列に動いているからと
いう理由であれ、いずれにしても時間というものの理解をさらに複雑にします。

3.4.1 並⾏システムにおける時間の性質
表⾯的には、時間というものは単純なもののように⾒えます。時間は、事象に
つけられる順番です。35 任意の事象 Aと Bがあれば、ありうるパターンは、A
が B の前に起こる、Aと B が同時に起こる、Aが B の後に起こる、のどれか
になります。例えば、銀⾏⼝座の例に戻って考えてみましょう。最初の時点で
100ドル⼊っている共同⼝座から Peterが 10ドルを引き出し、Paulが 25ドル
を引き出して、⼝座には 65ドルが残るとします。⼆回の引き出しの順番によ
って、残⾼の推移は $100 → $90 → $65 か $100 → $75 → $65 のどちらかに
なります。銀⾏システムのコンピュータ実装では、この変わっていく残⾼の推
移は、変数 balanceへの連続した代⼊としてモデル化できます。
しかし、状況が複雑になると、そのような⾒⽅には問題が出てきます。Peter

や Paulやほかの⼈たちが、世界中に分散した ATMのネットワークを通して、
同じ銀⾏⼝座にアクセスしていると考えてみましょう。実際の⼝座残⾼の推移
は、アクセスのタイミングや機械同⼠の通信の詳細に決定的に依存することに
なります。
このように、イベントの順番が決まっていないということは、並⾏システム

を設計するうえで深刻な問題になります。例えば、Peterと Paulによる引き出
しが、共通の変数 balanceを共有する⼆つの別々のプロセスとして実装されて
いるとします。各プロセスは、3.1.1節の⼿続きによって規定されるものです。

34実際のプロセッサの多くは、パイプライン (pipelining)という戦略によって、同時に
いくつかの演算を実⾏します。このテクニックはハードウェアの利⽤効率を⼤きく改善
するのですが、これは直列の命令列の実⾏スピードを上げるためだけに使われ、逐次的
プログラムのふるまいはそのままです。

35ケンブリッジの建物の壁に描かれた落書きからの引⽤:“時間とは、何もかもが同時に
起こらないようにするために発明された仕掛けである”

(define (withdraw amount)
(if (>= balance amount)

(begin
(set! balance (- balance amount))
balance)

"Insufficient funds"))

もし⼆つのプロセスが独⽴に動作するとすると、次のようなことが起こる可能
性があります。まず、Peterは残⾼を確認して、正当な⾦額を引き出そうとしま
す。しかし、Peterが残⾼を確認してから引き出しを終えるまでの間に、Paul
がお⾦を引き出してしまって、Peterの確認結果が無効になってしまいます。
もっとひどいことになる可能性もあります。それぞれの引き出しプロセス

の⼀部として、次の式が実⾏されるとします。
(set! balance (- balance amount))

この式は、三つのステップからなります。(1) 変数 balanceの値にアクセスし、
(2) 新しい残⾼を計算し、(3) balanceにこの新しい値を設定する、というもの
です。もし Peterと Paulの引き出し⼿続きがこの⽂を並⾏に実⾏したとする
と、⼆つの⼿続きが balanceにアクセスして新しい値を設定する順番が⼊れ違
いになってしまう可能性もあります。
図 3.29のタイミング図は、balanceが 100の状態から始まって、Peterが 10

引き出し、Paulが 25を引き出しているのに、balanceの最終的な値が 75とな
るという状況で、イベントの発⽣順を表したものです。図に⽰したように、こ
ういうおかしなことになった原因は、Paulの balanceへの 75の代⼊が、引き
算する balanceの値が 100であるという前提のもとで⾏われているというこ
とです。しかし、この前提は Peterが balanceを 90に変更したときに無効に
なっています。システム内のお⾦の総額が保たれていないので、これは銀⾏シ
ステムとしてはどうしようもない失敗です。⼆つの取引の前は、お⾦の総額は
100ドルでした。取引の後は、Peterが 10ドル、Paulが 25ドル、銀⾏が 75ド
ルを持っていることになっています。36

36⼆つの set!演算が同時に残⾼を変更しようとすると、もっとひどい失敗が起こる可
能性があります。この場合、メモリ上に残る実際のデータは、⼆つのプロセスが書き込ん
だ情報がでたらめに混ぜ合わされたものになってしまうかもしれません。ほとんどのコ
ンピュータでは、基本メモリ書き込み演算にはインターロックがあり、そのような同時ア
クセスを防いでいます。しかし、このような⼀⾒単純な保護の仕組みさえ、マルチプロセ
スコンピュータを設計する際には実装上の問題が出てきます。設計にあたっては、メモ
リアクセスの速度を上げるために別々のプロセッサの間でデータが複製 (“キャッシュ”)

Peter

Access balance: $100

new value: 100 –10 = 90

set! balance to $90

time

Bank Paul

$100

$90

$75

Access balance: $100

new value: 100 – 25 =75

set! balance to $75

図 3.29: ⼆回の引き出しで、イベントの順番が⼊れ違いにな
って最終的に間違った残⾼になってしまう様⼦を⽰したタ
イミング図

ここでの例は、複数のプロセスが共通の状態変数を共有するという⼀般的
な現象を⽰しています。これが⾯倒なことになるのは、⼆つ以上のプロセスが
同時に共有された状態を操作しようとする可能性があることによります。銀⾏
⼝座の例では、それぞれの取引の間は、それぞれの顧客がほかに顧客がいない
かのようにふるまうことができる必要があります。ある顧客が⼝座残⾼に依存
する形で残⾼を更新しようとしているとすると、変更のちょうどその前の瞬間
まで⼝座が⾃分の考えている通りの残⾼を持っているという前提で顧客が⾏動
できるようにしなければなりません。

並⾏プログラムの正しいふるまい
上の例は、並⾏プログラムに⼊り込みがちなバグの典型例を⽰しています。こ
の複雑性の根元には、異なるプロセスの間で共有される変数に対する代⼊があ
ります。ここまでですでに、set!を使うプログラムを書く際には気をつけなけ
ればならないということがわかりました。これは、計算の結果が代⼊の起こる
順番に依存するからです。37 並⾏プロセスでは、代⼊には特に気をつける必要
があります。これは、別々のプロセスが⾏う代⼊の順番はコントロールできる
とは限らないからです。そのような変更が (⼆⼈の預⾦者が共同⼝座にアクセ
スする場合のように)いくつか同時に起こる可能性があるなら、何らかの⽅法
でシステムの正常な動作を保証しなければなりません。例えば、共同⼝座から
の引き出しの場合であれば、⾦額が変わらないことを保証する必要があります。
並⾏プログラムを正常に動作させるためには、並⾏実⾏に何らかの制限を加え
ることが必要になるかもしれません。
考えられる制限のひとつとして、共有された状態を変更するようなどんな

⼆つの演算も同時に起こってはならないと決めてしまうことです。これは⾮常
に厳しい要求です。分散銀⾏システムの場合であれば、同時にはひとつの取引
しか⾏えないことを保証するようシステム設計者に要求することになります。
これは⾮効率的で、また保守的すぎます。図 3.30は、Peterと Paulが⼝座を共
有していて Paulはほかに個⼈⼝座も持っているという状況を⽰したものです。
図は、共有⼝座からの⼆回の引き出し (Peterが⼀回、Paulが⼀回)と、Paul
の個⼈⼝座への預⾦を⽰しています。38共有⼝座からの⼆回の引き出しは、並

されていても、プロセッサ間で⼀貫したメモリ内容の⾒え⽅を保つことを保証するため
に、精巧なキャッシュコヒーレンス (cache-coherence) プロトコルが必要となります。

373.1.3節の階乗プログラムは、単⼀の逐次処理の場合にもこの問題が起こることを⽰
したものです。

38それぞれの列は、Peterの財布、(Bank1内の)共同⼝座、Paulの財布、(Bank2内

$100$7 $5 $300

$0 $305

$305$25$65$17

$17 $90

W

W

D

time

Peter Bank1 Paul Bank2

図 3.30: 銀⾏ 1の共同⼝座と銀⾏ 2の個⼈⼝座に対する並
⾏な預け⼊れと引き出し

⾏であってはいけません (どちらも同じ⼝座にアクセス・更新しているからで
す)。また、Paulの預⾦と引き出しは並⾏であってはいけません (どちらも Paul
の財布にアクセス・更新を⾏っているからです)。しかし、Paulが⾃分の個⼈
⼝座に預け⼊れを⾏うのと、Peterが共有⼝座から引き出しを⾏うのは、許可
しても問題ないはずです。
並⾏性に対するもう少し緩い制限として、各プロセスが何らかの順番で逐

次的に実⾏された場合と同じ結果になることを保証するというものがありま
す。この要求には、⼆つの重要な側⾯があります。⼀つ⽬は、各プロセスが別
々に実⾏される必要はなく、あたかもそれらのプロセスが逐次的に実⾏された
の)Paulの個⼈⼝座の中⾝が、それぞれの引き出し (W)と預け⼊れ (D)の前後でどうな
っているかを⽰しています。Peterは Bank1から 10ドルを引き出しています。Paulは
5ドルを Bank2に預⾦し、次に Bank1から 25ドルを引き出しています。

かのような結果さえ出せばいいということです。図 3.30の例でいうと、銀⾏⼝
座システムを設計する⼈は、Paulの預け⼊れと Peterの引き出しが同時に起こ
るようにしても問題ないことになります。全体としての結果は、⼆つの操作が
逐次的に⾏われたのと同じことになるからです。⼆つ⽬は、並⾏プログラムの
実⾏結果には⼆つ以上の可能な “正解”がありうるということです。プログラ
ムの結果は、何らかの順序で逐次的に実⾏したものと同じになっていればよい
からです。例えば、Peterと Paulの共同⼝座の残⾼が最初は 100ドルで、Peter
が 40ドルを預け⼊れるのと並⾏に Paulが残⾼の半額を引き出すという場合に
ついて考えてみましょう。この場合、逐次実⾏では、残⾼は最終的に 70ドルか
90ドルのどちらかになります (練習問題 3.38参照)。39

並⾏プログラムを正しく動作させる要件として、これより弱いものもあり
ます。例えば、(物体内での熱の流れのような)拡散シミュレーションプログラ
ムを数多くのプロセスから構成されるものとして書く場合について考えてみま
す。それぞれのプロセスは⼩さな空間を表現し、値を並⾏に更新するものとし
ます。それぞれのプロセスは、⾃分⾃⾝の値と近傍の値の平均を取って、連続
して⾃分の値を変えていきます。このアルゴリズムは、演算の⾏われる順番と
は独⽴に、正しい解に収束します。共有される値を並⾏に使⽤する際に制限を
設ける必要はありません。

練習問題 3.38: Peter, Paul, Maryが共同⼝座を持っていて、最初
は 100ドル⼊っているとする。以下に⽰すコマンドによって、Peter
は 10ドルを預け⼊れ、Paulは 20ドルを引き出し、Maryは⼝座
残⾼の半額を引き出すということを並⾏的に⾏う。
Peter: (set! balance (+ balance 10))
Paul: (set! balance (- balance 20))
Mary: (set! balance (- balance (/ balance 2)))

a. これらの 3つの取引が完了した後で balanceが取りうる異な
る値をすべて列挙せよ。銀⾏システムは、三つのプロセスが
何らかの順序で逐次的に実⾏されるようにするとする。

b. もしシステムがプロセスの実⾏が⼊れ違いになることを許す
とすると、ほかにどのような値になる可能性があるだろうか。

39この考え⽅をより形式的に表現すると、並⾏プログラムは本質的に ⾮決定的 (non-
deterministic)であるということになります。つまり、並⾏プログラムは単⼀の値を持つ
関数として記述するのではなく、取りうる値の集合という形で結果を持つ関数として記
述することになります。4.3節では、⾮決定的計算を表現する⾔語について学びます。

図 3.29のようなタイミング図を描き、それらの値がどのよう
にして起こるか説明せよ。

3.4.2 並⾏性制御メカニズム
ここまで⾒てきたように、並⾏プロセスを扱ううえで難しいところは、別々の
プロセスのイベントの順番が⼊れ違いになることを考えないといけないことに
よるものです。例えば、⼆つのプロセスがあって、⽚⽅は (a, b, c)、もう⽚⽅
は (x, y, z)という、それぞれ順序つきの三つのイベントを持っているとします。
もし⼆つのプロセスが、実⾏が⼊れ違いになることに対する制限がない状態で
並⾏して動作するとすると、個別プロセス内での順序と⽭盾しないイベントの
順序として、20通りが考えられることになります。

(a,b,c,x,y,z) (a,x,b,y,c,z) (x,a,b,c,y,z) (x,a,y,z,b,c)
(a,b,x,c,y,z) (a,x,b,y,z,c) (x,a,b,y,c,z) (x,y,a,b,c,z)
(a,b,x,y,c,z) (a,x,y,b,c,z) (x,a,b,y,z,c) (x,y,a,b,z,c)
(a,b,x,y,z,c) (a,x,y,b,z,c) (x,a,y,b,c,z) (x,y,a,z,b,c)
(a,x,b,c,y,z) (a,x,y,z,b,c) (x,a,y,b,z,c) (x,y,z,a,b,c)

プログラマがこのシステムを設計するにあたっては、これら 20通りの順序の
効果を考え、それぞれのふるまいが受け⼊れられるものかどうかチェックしな
ければならなくなります。そのようなやり⽅は、プロセスとイベントの数が増
えるにつれて、あっという間に⼿に負えなくなります。
より現実的な並⾏システムの設計アプローチとしては、プログラムのふる

まいの正しさを保証できる形で並⾏プロセスの⼊れ違い⽅を制約できるような
⼀般的なメカニズムを考えるというものがあります。これまで、このことを⽬
的とした多くのメカニズムが開発されてきました。この節では、その中のひと
つである直列変換器 (serializer)について説明します。

共有状態へのアクセスの直列化
直列化というのは、次のような考え⽅を実装したものです。「プロセスは並⾏
して実⾏されるが、並⾏して実⾏できない⼿続きの集合がある」。より正確に
は、直列化では互いに区別される複数の⼿続きの集合を作り、それぞれの直列
化された集合の中では同時にひとつのプロセスしか実⾏されないようにすると
いうものです。集合内の⼿続きが実⾏されているときには、その集合内の何か

の⼿続きを実⾏しようとするプロセスは、⼀つ⽬の実⾏が終わるまで待たされ
ることになります。
直列化は共有変数へのアクセスの制御に使えます。例えば、共有変数の値

をその変数のそれまでの値に基づいて更新したいというとき、その変数のそれ
までの値に対するアクセスとその変数への新しい値の代⼊を同じ⼿続き内に置
くようにします。それから、その変数に値を割り当てるほかの⼿続きもすべて
同じ直列変換器にまとめることで、この⼿続きとそれらの⼿続きが同時に実⾏
されないようにします。こうすることで、変数へのアクセスとそれに対応する
値の割り当てとの間で変数の値が変更されないことが保証できます。

Schemeのシリアライザ
上のメカニズムをもう少し具体的に考えるために、parallel-executeという
⼿続きを持つ拡張 Schemeがあると仮定します。
(parallel-execute ⟨p1⟩ ⟨p2⟩ . . . ⟨pk⟩)

各 ⟨p⟩は、引数なしの⼿続きである必要があります。parallel-executeは、各
⟨p⟩に対して独⽴したプロセスを作り、それらのプロセスは ⟨p⟩を (引数なし
で)適⽤します。これらのプロセスは、すべて並⾏に実⾏されます。40

これの使い⽅の例として、以下のものについて考えてみましょう。
(define x 10)
(parallel-execute
(lambda () (set! x (* x x)))
(lambda () (set! x (+ x 1))))

これは、⼆つの並⾏プロセスを作ります。xに xかける xを設定する P1 と、x
に 1を⾜す P2 です。実⾏が終わった後には、P1 と P2 のイベントの⼊れ違い
具合によって、xは可能な五通りの値のうちのひとつになります。
101: P1 が xに 100を設定し、次に P2 が xを 101に増やす
121: P2 が xを 11に増やし、次に P1 が xを x * xに設定する
110: P1 が (* x x)を評価する際の xの値に対する⼆回のアクセスの間に

P2 が xを 10から 11に変える
40parallel-executeは標準 Schemeの⼀部ではありませんが、mit Schemeで実装で
きます。ここでの実装では、新しい並⾏プロセスはオリジナルの Schemeプロセスと並
⾏に実⾏できます。また、ここでの実装では、parallel-executeによって返される値は、
新しく作ったプロセスを停⽌させるのに使える特別なコントロールオブジェクトです。

11: P2 が xにアクセスし, それから P1 が xを 100に設定し、
それから P2 が xを設定する

100: P1 が xに (⼆回)アクセスし、それから P2 が xを 11に設定し、
それから P1 が xを設定する

直列化⼿続きを使うことによって、並⾏性に制約を加えることができます。直
列化⼿続きは、シリアライザ (serializer)によって作ります。シリアライザは、
make-serializerにより構築されます。この実装は後で⽰します。シリアライ
ザは⼿続きを引数に取り、元の⼿続きと同じようにふるまう直列化⼿続きを返
します。あるひとつのシリアライザに対する呼び出しは、すべて同じ集合に属
する直列化⼿続きを返します。
このため、次のものを実⾏すると、上の例とは異なり、xの⼆つの可能な値

である 101または 121しか返しません。
(define x 10)
(define s (make-serializer))
(parallel-execute
(s (lambda () (set! x (* x x))))
(s (lambda () (set! x (+ x 1)))))

ほかの可能性は排除されています。これは、P1 と P2 の実⾏が⼊れ違いになら
ないようになっているからです。

3.1.1節の make-account⼿続きの、預け⼊れと引き出しが直列化されたバ
ージョンを以下に⽰します。
(define (make-account balance)

(define (withdraw amount)
(if (>= balance amount)

(begin (set! balance (- balance amount))
balance)

"Insufficient funds"))
(define (deposit amount)

(set! balance (+ balance amount))
balance)

(let ((protected (make-serializer)))
(define (dispatch m)

(cond ((eq? m 'withdraw) (protected withdraw))
((eq? m 'deposit) (protected deposit))

((eq? m 'balance) balance)
(else (error "Unknown request: MAKE-ACCOUNT"

m))))
dispatch))

この実装では、⼆つのプロセスがひとつの⼝座に並⾏に預け⼊れをしたり引き
出しをしたりすることはできなくなっています。こうすることで、図 3.29で
図⽰したような、Paulが新しい値を計算するために残⾼にアクセスしてから
Paulが実際に代⼊を⾏うまでの間に Peterが⼝座残⾼を変えてしまうといった
エラーの原因は排除されます。⼀⽅、⼆つの⼝座はそれぞれ⾃分のシリアライ
ザを持っているので、別々の⼝座に対する預け⼊れと引き出しは並⾏に⾏うこ
とができます。

練習問題 3.39: 並⾏実⾏の結果として上で⽰した五つの可能な値
のうち、実⾏を次のように直列化したら、どれが残るだろうか。
(define x 10)
(define s (make-serializer))
(parallel-execute
(lambda () (set! x ((s (lambda () (* x x))))))
(s (lambda () (set! x (+ x 1)))))

練習問題 3.40: 以下のものを実⾏したときに xが取りうる値をす
べて列挙せよ。
(define x 10)
(parallel-execute (lambda () (set! x (* x x)))

(lambda () (set! x (* x x x))))

以下の直列化⼿続きを使うと、これらの可能性のうちどれが残る
だろうか。
(define x 10)
(define s (make-serializer))
(parallel-execute (s (lambda () (set! x (* x x))))

(s (lambda () (set! x (* x x x)))))

練習問題 3.41: Ben Bitdiddleは、次のように銀⾏⼝座を実装した
ほうがいいのではないかと⼼配している (変更箇所はコメントつ
きの⾏)。

(define (make-account balance)
(define (withdraw amount)

(if (>= balance amount)
(begin (set! balance

(- balance amount))
balance)

"Insufficient funds"))
(define (deposit amount)

(set! balance (+ balance amount))
balance)

(let ((protected (make-serializer)))
(define (dispatch m)

(cond ((eq? m 'withdraw) (protected withdraw))
((eq? m 'deposit) (protected deposit))
((eq? m 'balance)
((protected

(lambda () balance)))) ; 直列化
(else
(error "Unknown request: MAKE-ACCOUNT"

m))))
dispatch))

その理由は、銀⾏⼝座に対する⾮直列化アクセスを許していると、
異常なふるまいが起こるかもしれないということだ。その意⾒に
賛成するか。Benの⼼配を実証するようなシナリオは存在するだ
ろうか。
練習問題 3.42: Ben Bitdiddleは、withdrawと depositメッセー
ジが来るたびに新しい直列化⼿続きを作るのは時間の無駄だと提
案した。彼の意⾒は、make-accountを変更して、protectedに対
する呼び出しを dispatch⼿続きの外で⾏うようにしたらいいと
いうものだ。つまり、⼝座が引き出し⼿続きを要求されるたびに、
(⼝座作成と同時に作った)同じ直列化⼿続きを返すことになる。
(define (make-account balance)

(define (withdraw amount)
(if (>= balance amount)

(begin (set! balance (- balance amount))

balance)
"Insufficient funds"))

(define (deposit amount)
(set! balance (+ balance amount))
balance)

(let ((protected (make-serializer)))
(let ((protected-withdraw (protected withdraw))

(protected-deposit (protected deposit)))
(define (dispatch m)

(cond ((eq? m 'withdraw) protected-withdraw)
((eq? m 'deposit) protected-deposit)
((eq? m 'balance) balance)
(else
(error "Unknown request: MAKE-ACCOUNT"

m))))
dispatch)))

これは安全な変更だろうか。具体的には、これら⼆つのバージョン
の make-accountによって許される並⾏性に違いはあるだろうか。

複数の共有リソースを使う際の複雑性
シリアライザは、並⾏プログラムの複雑性を隔離し、複雑性を注意深く、また
(おそらくは)正しく扱うことができるように、強⼒な抽象化を提供しています。
しかし、共有リソースが (ひとつの銀⾏⼝座のように)ひとつしかない場合に
はシリアライザを使うのは⽐較的素直にうまくいくのですが、共有リソースが
複数ある場合には、並⾏プログラミングは落とし⽳の多い難しいものになるこ
とがあります。
起こりうる困難のひとつを⽰すため、⼆つの銀⾏⼝座の残⾼を交換したい

という例について考えます。この場合、それぞれの⼝座にアクセスして残⾼を
調べ、残⾼同⼠の差を計算し、その差の分だけ⽚⽅の⼝座から引き出し、もう
ひとつの⼝座に預け⼊れることになります。これは、次のように実装できます。
41

(define (exchange account1 account2)

41depositメッセージが負の⾦額を受け⼊れるということを利⽤して exchangeを簡単
にしています (これはこの銀⾏システムの深刻なバグです！)。

(let ((difference (- (account1 'balance)
(account2 'balance))))

((account1 'withdraw) difference)
((account2 'deposit) difference)))

この⼿続きは、交換をしようとしているのがひとつのプロセスだけの場合
にしかうまくいきません。ここで、Peterと Paulの⼆⼈が両⽅とも⼝座 a1, a2,
a3にアクセスし、Peterが a1と a2を交換する間に Paulが並⾏に a1と a3を
交換するという場合について考えてみましょう。個別⼝座に対する預け⼊れと
引き出しが (この節の上で⽰した make-account⼿続きのように)直列化されて
いたとしても、exchangeはやはり不正確な結果になる可能性があります。例
えば、Peterが a1と a2の残⾼の差を求めるとき、Peterが交換を終える前に
Paulが a1の残⾼を変更してしまうかもしれません。42 正しいふるまいをさせ
るためには、交換が終わるまでの間ずっと、どちらの⼝座からもほかの並⾏ア
クセスを排除するように配慮する必要があります。
これを実現する⽅法のひとつとして、両⽅の⼝座のシリアライザを使って

exchange⼿続き全体を直列化するというものがあります。そのためには、⼝座
のシリアライザにアクセスする準備をします。シリアライザを露出させること
で、銀⾏⼝座オブジェクトのモジュール性を意図的に破っていることに注意し
てください。make-accountの以下のバージョンは、3.1.1節の元のバージョン
と同じですが、balance変数を守るためにシリアライザが提供されていること
と、そのシリアライザがメッセージパッシングによってエクスポートされてい
ることが違います。
(define (make-account-and-serializer balance)

(define (withdraw amount)
(if (>= balance amount)

(begin (set! balance (- balance amount))
balance)

"Insufficient funds"))
(define (deposit amount)

(set! balance (+ balance amount))
balance)

42⼝座残⾼が最初 10ドル, 20ドル, 30ドルだったとすると、並⾏の交換を何回⾏って
も、順番はともかく、残⾼は 10ドル, 20ドル, 30ドルのままでなければなりません。個
別⼝座に対する預け⼊れを直列化するだけでは、このことを保証するには⼗分ではあり
ません。練習問題 3.43参照。

(let ((balance-serializer (make-serializer)))
(define (dispatch m)

(cond ((eq? m 'withdraw) withdraw)
((eq? m 'deposit) deposit)
((eq? m 'balance) balance)
((eq? m 'serializer) balance-serializer)
(else (error "Unknown request: MAKE-ACCOUNT" m))))

dispatch))

これを使って、預け⼊れと引き出しを直列化して⾏うことができます。しかし、
前の直列化⼝座とは違って、直列化を明⽰的に管理するのは銀⾏⼝座オブジェ
クトのそれぞれのユーザーの責任になっています。以下はその⼀例です。43

(define (deposit account amount)
(let ((s (account 'serializer))

(d (account 'deposit)))
((s d) amount)))

このようにシリアライザをエクスポートすることによって柔軟性が得られ、交
換プログラムを直列化したものを実装できるようになります。単純に、両⽅の
⼝座のシリアライザによって元の exchange⼿続きを直列化するだけです。
(define (serialized-exchange account1 account2)

(let ((serializer1 (account1 'serializer))
(serializer2 (account2 'serializer)))

((serializer1 (serializer2 exchange))
account1
account2)))

練習問題 3.43: 三つの⼝座の残⾼が最初は 10ドル、20ドル、30ド
ルで、それから複数のプロセスが実⾏されて⼝座の残⾼を交換す
るとする。プロセスが逐次的に実⾏されるとしたら、並⾏の残⾼
交換を何回⾏ったとしても、⼝座残⾼は何らかの順番で 10ドル、
20ドル、30ドルであるということを論証せよ。この章の最初のバ
ージョンの⼝座交換プログラムを使って残⾼交換を実装するとこ
の条件が破られることもあるということを、図 3.29のようなタイ

43練習問題 3.45では、なぜ預け⼊れと引き出しが⼝座によって⾃動的に直列化されな
いようになったのかを調査します。

ミング図を描いて⽰せ。⼀⽅、この exchangeプログラムを使った
としても、⼝座残⾼の合計は保存されることを論証せよ。個別⼝
座の取引を直列化しなければこの前提まで破られるということを
タイミング図を描いて⽰せ。
練習問題 3.44: ある⼝座から別の⼝座に振込をするという問題に
ついて考える。Ben Bitdiddleは、たとえ複数の⼈々が複数の⼝座
間で並⾏にお⾦の振込をするという場合であっても、以下の⼿続
きによって達成できると主張している。この⼿続きでは、例えば
上で出てきたバージョンの make-accountのように、預け⼊れと引
き出しの取引を直列化する⼝座側の何らかのメカニズムを使う。
(define (transfer from-account to-account amount)

((from-account 'withdraw) amount)
((to-account 'deposit) amount))

Louis Reasonerは、これには問題があり、残⾼交換問題を扱う際に
必要となったような、より洗練された⼿法を使う必要があると主
張する。Louisは正しいだろうか。もし正しくないとしたら、振込
問題と残⾼交換問題の本質的な違いは何だろうか (from-account
の残⾼は amount以上あると考える)。
練習問題 3.45: Louis Reasonerは、預け⼊れと引き出しが⾃動的
に直列化されないようになったことで、この銀⾏⼝座システム
が必要以上に複雑になり、またエラーが起こりやすくなったと考
えている。彼は、make-account-and-serializerが (serialized-
exchangeのような⼿続きで使えるように)シリアライザをエクス
ポートするだけでなく、その上で make-accountのように⼝座と
預け⼊れを直列化するためにもそれを使うようにしたほうがいい
と提案した。彼は、⼝座を次のように再定義することを提案して
いる。
(define (make-account-and-serializer balance)

(define (withdraw amount)
(if (>= balance amount)

(begin (set! balance (- balance amount)) balance)
"Insufficient funds"))

(define (deposit amount)
(set! balance (+ balance amount)) balance)

(let ((balance-serializer (make-serializer)))

(define (dispatch m)
(cond ((eq? m 'withdraw) (balance-serializer withdraw))

((eq? m 'deposit) (balance-serializer deposit))
((eq? m 'balance) balance)
((eq? m 'serializer) balance-serializer)
(else (error "Unknown request: MAKE-ACCOUNT" m))))

dispatch))

すると、預け⼊れは元の make-accountと同じように扱われること
になる。
(define (deposit account amount)

((account 'deposit) amount))

Louisの考えのどこがおかしいのか説明せよ。具体的には、serialized-
exchangeが呼ばれたときにどうなるか考えよ。

シリアライザの実装
シリアライザは、ミューテックス (mutex)と呼ばれる、より基本的な同期メカ
ニズムによって実装します。ミューテックスは、獲得 (acquire)と解放 (release)
という⼆つの演算をサポートするオブジェクトです。ミューテックスが⼀度獲
得されると、そのミューテックスに対するほかの獲得演算は、そのミューテッ
クスが解放されるまで続⾏できません。44 ここでの実装では、それぞれのシリ
アライザは関連づけられたミューテックスをひとつ持っています。⼿続き pが
与えられると、シリアライザはミューテックスを獲得し、pを実⾏し、それか
らmutexを解放するという⼿続きを返します。これによって、このシリアライ
ザによって⽣成された⼿続きのうちひとつしか同時に実⾏されないことを保証
します。これこそが、保証しなければならない直列化の特性です。

44“mutex”という⽤語は相互排除 (mutual exclusion) の短縮形です。並⾏プロセスが
リソースを安全に共有できる仕組みを準備するという⼀般的な問題は、相互排除問題と
呼ばれます。ここでのミューテックスは、セマフォ (semaphore) メカニズムをちょっと
変えて単純にしたものです (練習問題 3.47参照)。セマフォというメカニズムは、アイン
トホーフェン技術⼤学で開発された “THE”マルチプログラミングシステムで導⼊され
たものです。“THE”という名前は、この⼤学のオランダ語のイニシャルに由来します
(Dijkstra 1968a)。獲得と解法の演算は、元は Pと Vと呼ばれていました。これらは鉄
道で使われる腕⽊信号との関連で、オランダ語の単語 passeren(渡す)と vrijgeven(解放
する)から来ています。ダイクストラの有名な解説 (Dijkstra 1968b)は、明確に並⾏性制
御の問題を提⽰した最初期のもののひとつで、セマフォを使ってさまざまな並⾏性の問
題を扱う⽅法を⽰しています。

(define (make-serializer)
(let ((mutex (make-mutex)))

(lambda (p)
(define (serialized-p . args)

(mutex 'acquire)
(let ((val (apply p args)))

(mutex 'release)
val))

serialized-p)))

ミューテックスは、真または偽の値を持つことができる可変オブジェクト (こ
こでは 1要素のリストを使い、それをセル (cell)と呼ぶことにします)です。値
が偽のときはミューテックスは獲得可能です。値が真のときはミューテックス
は使⽤中で、そのミューテックスを獲得しようとするプロセスは待たされるこ
とになります。
ここでのミューテックスのコンストラクタ make-mutexは、まずセルの中

⾝を偽に初期化します。そのミューテックスを獲得するには、セルを確認しま
す。もしミューテックスが使⽤可能であれば、セルの中⾝を真にして続⾏しま
す。そうでなければ、ミューテックスが使⽤可能になるまで、ループしながら
何度も獲得を試みます。45 ミューテックスを解放するには、セルの中⾝に偽を
設定します。
(define (make-mutex)

(let ((cell (list false)))
(define (the-mutex m)

(cond ((eq? m 'acquire)
(if (test-and-set! cell)

(the-mutex 'acquire))) ; retry
((eq? m 'release) (clear! cell))))

the-mutex))
(define (clear! cell) (set-car! cell false))

45ほとんどの時分割 OSでは、ミューテックスでブロックされたプロセスは、上記の
ように “ビジーウェイト”で時間を無駄にすることはありません。そうする代わりに、シ
ステムは最初のプロセスの待ち時間の間にほかのプロセスが動くようにスケジュールし、
ブロックされたプロセスはミューテックスが使⽤可能になったときに起こされることに
なります。

test-and-set!は、セルをテストし、テスト結果を返します。それに加えて、
もしテストが偽であれば、test-and-set!は偽を返す前にセルの中⾝に真を設
定します。このふるまいは、次のような⼿続きとして表現できます。
(define (test-and-set! cell)

(if (car cell) true (begin (set-car! cell true) false)))

しかし、この test-and-set!の実装は現状では⼗分ではありません。ここに⾮
常に重要な難しい問題があります。これは並⾏性制御がシステムに⼊り込むこ
とになる根本的なところなのですが、それは test-and-set!演算はアトミック
(atomically)に実⾏されなければいけないということです。これは、あるプロ
セスがセルをテストし偽だと知ったら、ほかのプロセスがそのセルをテストす
るより前に、そのセルの内容は実際に真に設定されるよう保証しなければなら
ないということを意味します。この保証をしなければ、図 3.29で銀⾏⼝座が失
敗したのと同じように、ミューテックスが失敗することになります (練習問題
3.46参照)。

test-and-set!の実際の実装は、システムが並⾏プロセスをどのように実
⾏しているかに依存します。例えば、並⾏プロセスを動かすのに、逐次的なプ
ロセッサ上で時分割メカニズムを使い、各プロセスに短い時間を割り当てて動
かしたのちに割り込みをして次のプロセスに処理を移すというやり⽅でプロ
セスを巡回するということをしているかもしれません。その場合、test-and-
set!はテストと値の設定の間は時分割を停⽌するということによってうまく
いきます。46 また、マルチプロセスコンピュータはアトミック演算を直接ハー
ドウェアでサポートする命令を提供します。47

46シングルプロセッサ向けの mit Schemeは時分割モデルを使うので、test-and-set!
は以下のように実装できます。
(define (test-and-set! cell)

(without-interrupts
(lambda ()

(if (car cell)
true
(begin (set-car! cell true)

false)))))

without-interruptsは、引数の⼿続きが実⾏されている間、時分割割り込みを無効にす
るものです。

47そのような命令には、test-and-set, test-and-clear, swap, compare-and-exchange,
load-reserve, store-conditionalなど、さまざまなものがあります。これらの設計は、そ

練習問題 3.46: テキストで⽰したように、test-and-set!を通常
の⼿続きを使って実装し、演算をアトミックにしようとはしない
と仮定する。図 3.29のようなタイミング図を描き、このミューテ
ックスの実装が、⼆つのプロセスが同時にミューテックスを獲得
するのを許してしまうという失敗を犯すことになるのはどのよう
な場合か⽰せ。
練習問題 3.47: (サイズ nの)セマフォとは、ミューテックスの⼀
般化である。ミューテックスと同じように、セマフォは獲得と解
放の演算をサポートするが、並⾏に nプロセスまで獲得できると
いう意味でより⼀般的になっている。それ以上のプロセスがセマ
フォを獲得しようとすると、解放演算を待つ必要がある。セマフ
ォの実装を、以下それぞれの条件で⾏え。

a. ミューテックスを使う
b. アトミックな test-and-set!演算を⽤いる

デッドロック
シリアライザの実装⽅法について学んだため、今度は上記の serialized-
exchangeを使っても⼝座の残⾼交換にはまだ問題があるということを⾒てい
きます。ここで、Peterが a1と a2の残⾼を交換しようとしたとき、Paulが並
⾏に a2と a1の残⾼を交換しようとしていると想像してみましょう。Peterの
プロセスは、a1を保護する直列化⼿続きに⼊った時点に達し、そのすぐ後に、
Paulのプロセスが a2を保護する直列化⼿続きに⼊ったとします。このとき、
a2を保護する直列化⼿続きから Paulが抜けるまで、Peterは進む (a2を保護
の機械のプロセッサ-メモリ間インターフェイスに注意深く合わせたものでなければいけ
ません。ここで起こる問題のひとつとして、そのような命令を使って⼆つのプロセスが
ちょうど同じ時間に同じリソースを獲得しようとした場合に何が起こるかを決めるとい
うことがあります。この場合には、どちらのプロセスが制御を得るかについての決断を
するメカニズムが必要になります。そのようなメカニズムは、アービタ (arbiter)と呼ば
れます。アービタは通常、最終的には何らかのハードウェアデバイスに⾏き着きます。残
念なことに、アービタの決定に任意の⻑い時間を許さない限り、100%の確率でちゃんと
動作する公平なアービタを構築することは物理的に不可能だと証明可能です。ここでの
基本的な現象について最初に触れているのは、アリストテレスの天体論に対する 14世紀
のフランス⼈哲学者ジャン・ビュリダンの注釈です。ビュリダンは、等しく魅⼒的な⼆
つの⾷べ物の間に置かれた完全に理性的な⽝は、どちらに最初に⾏くか決めることがで
きないため飢え死にすると主張しました。

する直列化⼿続きに⼊る)ことができません。同じように、a1を守る直列化⼿
続きから Peterが抜けるまで、Paulは進むことができません。どちらのプロセ
スも、お互いを待って、永遠に⽴ち往⽣することになります。この状況はデッ
ドロック (deadlock)と呼ばれます。複数の共有リソースに対して並⾏なアクセ
スを提供するシステムでは、常にデッドロックに気をつける必要があります。
この状況でデッドロックを避けるひとつの⽅法として、それぞれの⼝座に

固有の ID番号を振り、プロセスが⼿続きに⼊る際に最⼩の番号を持つ⼝座を
最初に保護するように serialized-exchangeを書き換えるというものがあり
ます。この⼿法は交換問題にはうまくいくのですが、さらに洗練されたデッド
ロック回避テクニックが必要になる状況や、デッドロックがまったく防げない
という状況もあります (練習問題 3.48、練習問題 3.49参照) 48

練習問題 3.48: 上で説明したデッドロック回避⼿法 (⼝座に番号を
振り、各プロセスは⼩さな番号の⼝座から獲得を試みる) によっ
て、残⾼交換問題でのデッドロックが防げるのはなぜか、詳細に解
説せよ。serialized-exchangeを書き直し、この考え⽅を取り⼊
れよ (make-accountも変更して、各⼝座が番号つきで作られ、適
切なメッセージを送ることによってその番号にアクセスできるよ
うにする必要がある)。
練習問題 3.49: 上で説明したデッドロック回避メカニズムがうま
くいかない場合のシナリオを⽰せ (ヒント：残⾼交換問題では、各
プロセスは⾃分がどの⼝座にアクセスする必要があるか事前に知
っている。プロセスが、どの追加リソースが必要になるかわから
ない状態で共有リソースにアクセスする必要があるという状況を
考えよ)。

並⾏性、時間、通信
ここまで、並⾏システムのプログラミングでは、異なるプロセスが共有状態に
アクセスする際にイベントの順番を制御することが必要となるということを⾒
てきました。さらに、シリアライザをうまく使うことによってその制御を⾏う

48共有リソースに番号を振り、順番に獲得するという⼀般的なデッドロック回避技術
は、Havender (1968)によります。デッドロックが防げない状況では、デッドロック回
復 (deadlock-recovery)⼿法が必要となります。そのような⼿法では、プロセスにデッド
ロック状態から “⾝を引かせて”、再試⾏させます。デッドロック回復メカニズムはデ
ータベース管理システムで広く使⽤されています。これについては、Gray and Reuter
1993で詳しく取り上げられています。

⽅法についても⾒ました。しかし、並⾏性の問題はこれよりも深いところにあ
ります。根本的な観点から⾒ると、“共有状態”というものが何を意味するかが
常にはっきりしているとは限らないからです。

test-and-set!のようなメカニズムは、任意の時間にグローバルな共有フ
ラグを調べることをプロセスに要求します。これは、現代の⾼速なプロセッサ
で実装するには問題が多く、効率がよくありません。そのようなプロセッサで
は、パイプラインやキャッシュメモリといった最適化技術のために、すべての
瞬間にメモリの内容が⼀貫した状態にあるとは限らないからです。そのため、
最近のマルチプロセスシステムでは、シリアライザという枠組みは新しい並⾏
性制御法に取って代わられつつあります。49

共有状態の難しい側⾯は、⼤きな分散システムでも問題になります。例え
ば、分散銀⾏システムで、個別の⽀店が残⾼の局所的な値を保持していて、こ
れらの値をほかの⽀店で保持されているものと定期的に⽐較するという場合に
ついて想像してみましょう。そのようなシステムでは、“⼝座残⾼”の値は同期
直後以外では不確定になります。もし Peterが、Paulとの共同⼝座にお⾦を預
けたとしたら、⼝座残⾼が変わったのはいつということになるでしょうか。地
域⽀店が残⾼を変更したときでしょうか、それとも同期の後でしょうか。また、
もし Paulが別の⽀店から⼝座にアクセスしたとすると、ふるまいが “正しい”
といえるようなものにするためには、銀⾏システムにどのような制約を加える
のが妥当でしょうか。正しさについて考えるうえで意味があるのは、Peterと
Paulによって個別に観測されるふるまいと、同期直後の⼝座の “状態”のみで
す。“本当の”⼝座残⾼がいくらかとか、同期の間のイベントの順番というもの
は、重要でなかったり、意味がなかったりするかもしれません。50

ここでの基本的な現象は、異なるプロセスの同期や、共有状態の設定や、イ
ベントの順番を割り当てるといったことには、プロセス間の通信が必要となる
ということです。本質的に、並⾏性制御における時間という概念は、通信と密

49このような直列化の代替⼿段のひとつとして、バリア同期 (barrier synchronization)
と呼ばれるものがあります。プログラマは並⾏プロセスを好き勝⼿に実⾏させますが、あ
る⼀定の同期ポイント (“バリア”)を設定しておき、すべてのプロセスがそのバリアに着
くまで先に進めないようにするというものです。現代のプロセッサは、⼀貫性が必要とな
る部分にプログラマが同期ポイントを設定できるよう、機械語命令を提供しています。例
えば、PowerPCはこの⽬的のために sync(同期)と eieio(Enforced In-order Execution
of Input/Output)と呼ばれる⼆つの命令を含んでいます。

50これは奇妙な⾒⽅のように思えるかもしれません。しかし、このように動くシステム
は存在します。例えば、クレジットカードの⼝座に対する国際課⾦は、通常は国ごとに
精算され、別々の国で⾏われた課⾦は定期的に調整されます。このため、⼝座残⾼が国
によって異なるということがありえます。

接に結びついている必要があります。51 ⾯⽩いことに、相対論でも時間と通信
との間には似たようなつながりがあります。相対論では、光速 (イベントの同
期に使える最も⾼速な信号)は、時間と空間に関する基本的な定数です。計算
モデルの中での時間と状態を扱う際に遭遇する複雑性は、実際のところ、物理
的宇宙の基本的な複雑性を反映しているのかもしれません。

3.5 ストリーム
ここまでで、モデル化の道具としての代⼊について⼗分理解し、また代⼊が引
き起こす複雑な問題も認識できました。このあたりで、私たちがもし別の取り
組み⽅をしていたら、これらの問題のいくつかは避けられたのではないかとい
うことを考えてみてもいいころです。この節では、状態をモデル化する別のア
プローチについて検討します。このアプローチは、ストリーム (stream)と呼ば
れるデータ構造に基づくものです。ここから先で⾒ていくように、状態をモデ
ル化する際に⽣じる複雑性の⼀部をストリームによって軽減できます。
⼀歩下がって、この複雑性というものがどこから来ているのか再検討して

みましょう。現実世界の現象をモデル化しようとする中で、⾒たところもっと
もな決定をいくつかしてきました。局所状態を持つ現実世界のオブジェクトを
局所変数を持つ計算オブジェクトによってモデル化し、現実世界の時間の変化
をコンピュータ上の時間変化と同⼀視し、コンピュータ上のモデルオブジェク
トの状態の経時変化はモデルオブジェクトの局所変数に対する代⼊として実装
しました。
ほかのアプローチはないのでしょうか。モデル対象の世界での時間とコン

ピュータ上の時間を同⼀視することは避けられないのでしょうか。変化する世
界での現象をモデル化するには、時間とともにモデルを変化させるしかないの
でしょうか。この問題について、数学関数を使って考えてみましょう。量 xの
時間によって変化するふるまいは、時間の関数 x(t)として記述できます。もし
時刻ごとの xに注⽬するなら、それを変化する量として捉えることになります。
しかし、全時間を通しての値の履歴に注⽬するなら、変化については重要視し
ないことになります。関数そのものは変化しないことになるからです。52

51分散システムでは、この視点はLamport (1978)によって深く研究されています。彼
は、通信によって “グローバル時計”を作り、それを使って分散システムでイベントの順
番を確⽴する⽅法を⽰しました。

52物理学では、運動について考える⼿段として粒⼦の “世界線”を導⼊することによっ
て、この⾒⽅を採⽤することがあります。また、2.2.3節では、信号処理システムに関し

もし時間を離散ステップで測るのであれば、時間関数は (無限かもしれな
い)列としてモデル化できます。この節では、モデル化対象システムの時間履
歴を表現する列によって変化をモデル化する⽅法について⾒ていきます。これ
を達成するために、ストリーム (stream)と呼ばれる新しいデータ構造を導⼊し
ます。抽象的な視点から⾒ると、ストリームはただの列です。しかし、ストリ
ームを (2.2.1節のように)素直にリストとして実装したものでは、ストリーム
処理の本来の⼒が⾒えてきません。その代わりに、ここでは遅延評価 (delayed
evaluation)というテクニックを導⼊します。遅延評価を使うことによって、⾮
常に⼤きな (無限でもありうる)列をストリームして表現することが可能にな
ります。
ストリーム処理によって、状態のあるシステムを代⼊や可変データを使わ

ずにモデル化できるようになります。このことは、論理的にも実⽤的にも⼤き
な意味があります。代⼊の導⼊につきもののマイナス点を避けてモデルを構築
できるからです。⼀⽅、ストリームという枠組みにはそれはそれで難しいとこ
ろがあります。どちらのモデル化テクニックが、よりモジュール化され、より
メンテナンスしやすいシステムの実現につながるかということは、未解決の問
題です。

3.5.1 ストリーム＝遅延リスト
2.2.3節で⾒てきたように、列はプログラムモジュールを組み合わせるための標
準インターフェイスとして使うことができます。また、map, filter, accumulate
のような列操作の強⼒な抽象化について定式化してきました。これらは、簡潔
でエレガントに広い範囲の演算を表すことができるものです。
残念ながら、列をリストとして表すと、エレガントさのために計算に必要

な時間と空間の効率性を⼤幅に犠牲にすることになります。列の操作をリスト
の変形として表すと、プログラムは処理の各ステップで (⾮常に⼤きいかもし
れない)データ構造を構築し、コピーしないといけないことになります。
このことを確認するために、ある区間のすべての素数の和を求めるプログ

ラム⼆つを⽐較してみましょう。⼀つ⽬のプログラムは、標準的な繰り返しス
タイルで書かれています。53

て考えるうえではこれが⾃然な⽅法であると述べました。ストリームを信号処理に適⽤
することについては、3.5.3節で検討します。

53(1.2.6節で出てきたような)素数性をテストする述語 prime?はすでにあるとします。

(define (sum-primes a b)
(define (iter count accum)

(cond ((> count b) accum)
((prime? count)

(iter (+ count 1) (+ count accum)))
(else (iter (+ count 1) accum))))

(iter a 0))

⼆つ⽬のプログラムは、同じ計算を2.2.3節の列演算を使って⾏います。
(define (sum-primes a b)

(accumulate +
0
(filter prime?

(enumerate-interval a b))))

計算を実⾏する際に、⼀つ⽬のプログラムは、集積している和だけ格納すれ
ば⼗分です。それに対して、⼆つ⽬のプログラムのフィルタは、enumerate-
intervalが区間内の数の完全なリストを構築し終わるまで、まったくテスト
を⾏うことができません。フィルタは別のリストを⽣成し、それが accumulate
に渡され、その後に和に畳み込まれます。そのような⼤きな中間ストレージは
⼀つ⽬のプログラムでは必要ありません。⼀つ⽬のプログラムは、区間を順次
列挙し、素数を⽣成しながら和に⾜していくと考えることができます。
リストを使うことによる⾮効率性は、次のような例で痛切に明らかになり

ます。10,000 から 1,000,000 の区間に含まれる⼆番⽬の素数を計算するのに、
列のパラダイムを使って、次の式を評価することによって⾏うような場合です。
(car (cdr (filter prime?

(enumerate-interval 10000 1000000))))

この式は、確かに⼆番⽬の素数を⾒つけます。しかし、計算のオーバーヘッド
はものすごいものになります。約 100万の整数のリストを構築し、各要素の素
数性をテストしてこのリストをフィルタし、それからほとんどすべての結果を
無視することになってしまいます。より伝統的なプログラミングスタイルでは、
数え上げとフィルタをひとつにまとめ、⼆つ⽬の素数に到達した時点で停⽌す
ることになるでしょう。
ストリームは、列をリストとして操作するコストを発⽣させることなしに

列操作を使えるようにする巧妙な考え⽅です。ストリームを使うと、⼆つの世
界のいいとこ取りができます。列操作としてプログラムをエレガントに定式化

しながら、順次計算の効率性も得ることができるのです。基本的な考え⽅は、
ストリームを部分的しか構築しないで、構築した⼀部をそのストリームを消費
するプログラムに渡すというものです。消費プログラムがストリームのまだ構
築していない部分にアクセスしようとしたら、ストリームは⾃動的に必要な部
分だけを構築し、ストリーム全体が存在するという幻想を維持します。⾔い換
えると、プログラムはあたかも完全な列を扱っているかのように書くのですが、
ストリームの実装は、ストリームの構築とストリームの使⽤が⾃動的かつ透過
的に組み合わせられるように設計します。
表⾯的には、ストリームはただのリストで、それを⾛査する⼿続きの名前

が違うだけです。ストリームにはコンストラクタ cons-streamと⼆つのセレク
タ stream-car, stream-cdrがあり、セレクタは以下の制約を満たしています。
(stream-car (cons-stream x y)) = x
(stream-cdr (cons-stream x y)) = y

さらに、識別可能なオブジェクト the-empty-streamがあります。これは決し
て cons-stream演算の結果になることがなく、述語 stream-null?によって識
別可能です。54 これによって、リストを作ったり使ったりするのとまったく同
じようにストリームを作ったり使ったりすることができ、列という形になって
いる集約データを表現できます。具体的には、第 2章で学んだ list-ref, map,
for-eachのようなリスト演算について、ストリーム版を構築することが可能
です。55

(define (stream-ref s n)
(if (= n 0)

(stream-car s)
(stream-ref (stream-cdr s) (- n 1))))

(define (stream-map proc s)
(if (stream-null? s)

the-empty-stream
(cons-stream (proc (stream-car s))

54mitの実装では、the-empty-streamは空のリスト'()と同じもので、stream-null?
は null?と同じものとなっています。

55ここは気になるところだと思います。ストリームとリストに対してこのような似通
った⼿続きを定義しているということは、何か隠れた抽象化を⾒逃しているということ
です。残念ながら、この抽象化を利⽤するためには、評価プロセスに対する細かい制御が
必要なのですが、それは現時点ではまだできません。この点については、3.5.4節の末尾
で詳しく検討します。4.2節では、リストとストリームを統合する枠組みを開発します。

(stream-map proc (stream-cdr s)))))
(define (stream-for-each proc s)

(if (stream-null? s)
'done
(begin (proc (stream-car s))

(stream-for-each proc (stream-cdr s)))))

stream-for-eachはストリームを⾒るのに便利です。
(define (display-stream s)

(stream-for-each display-line s))
(define (display-line x) (newline) (display x))

ストリームが⾃動的かつ透過的にストリームの構築と使⽤を組み合わせるよう
実装するために、ストリームの cdrは、ストリームが cons-streamによって構
築されたときでなく、stream-cdr⼿続きによってアクセスされたときに評価
されるようにします。ここでの実装の選択は、2.1.2節で有理数について検討
したときと似ています。有理数を実装する際には、分⺟と分⼦を既約にするの
を構築時にするか選択時にするかを選ぶことができるということを⾒てきまし
た。どちらの有理数の実装も結果として同じデータ抽象化になりますが、どち
らを選ぶかによって効率に影響があります。ストリームと通常のリストの間に
も、似たような関係があります。データ抽象化としては、ストリームはリスト
と同じです。違うのは要素が評価されるタイミングです。通常のリストでは、
carも cdrも構築時に評価されます。ストリームでは、cdrは選択時に評価さ
れます。
ここでは、delay という特殊形式に基づいてストリームを実装します。

(delay <exp>)を評価しても、式 ⟨exp⟩は評価されず、代わりにいわゆる遅延
オブジェクト (delayed object)を返します。これは、ある将来の時点で ⟨exp⟩を
評価するという “約束” と考えることができます。delayと対になるものとし
て、forceという⼿続きがあります。これは、遅延オブジェクトを引数として
取り、評価を実⾏するというものです。つまり、delayに約束の遂⾏を迫ると
いうことになります。delayと forceの実装⽅法については後ほど⾒ていくと
して、まずはこれらを使ってストリームを構築してみましょう。
cons-streamは、次の形が
(cons-stream ⟨a⟩ ⟨b⟩)

以下と等価になるよう定義された特殊形式です。
(cons ⟨a⟩ (delay ⟨b⟩))

つまり、ストリームを構築するのにはペアを使うということです。しかし、ス
トリームの残りの値を cdrに⼊れる代わりに、要求されることがあれば残りを
計算するという約束を⼊れることにします。これで、stream-carと stream-
cdrが次のように⼿続きとして定義できるようになります。
(define (stream-car stream) (car stream))
(define (stream-cdr stream) (force (cdr stream)))

stream-carはペアの carを選択します。stream-cdrはペアの cdrを選択し、
そこに置かれた遅延表現を評価し、ストリームの残りを取得します。56

ストリーム実装の動作
この実装のふるまいを⾒るために、上で⾒た “とんでもない”素数計算をスト
リームによって定式化し直したものを解析してみましょう。
(stream-car
(stream-cdr
(stream-filter prime?

(stream-enumerate-interval
10000 1000000))))

これが実際に効率的に動くということを⾒ていきます。
まず、stream-enumerate-intervalを 10,000と 1,000,000という引数で呼

び出します。stream-enumerate-intervalは、enumerate-interval (2.2.3節)
のストリーム版です。
(define (stream-enumerate-interval low high)

(if (> low high)
the-empty-stream
(cons-stream
low
(stream-enumerate-interval (+ low 1) high))))

56stream-carと stream-cdrは⼿続きとして定義できるのですが、cons-streamは特
殊形式である必要があります。cons-streamが⼿続きだったとすると、私たちの評価モ
デルによると、(cons-stream <a> <b>)を評価すると⾃動的に ⟨b⟩が評価されることに
なります。これは避けたいところです。同じ理由から、delayも特殊形式でなければなり
ません。しかし、forceは通常の⼿続きでも⼤丈夫です。

つまり、stream-enumerate-intervalに返される結果は、cons-streamによっ
て作られた次のようなものになります。57

(cons 10000
(delay (stream-enumerate-interval 10001 1000000)))

つまり、stream-enumerate-intervalはストリームを返すのですが、そのスト
リームの carには 10,000が⼊っていて、cdrには、要求されたら区間の数え上
げを続けるという約束が⼊っていることになります。次に、このストリームに
フィルタをかけて素数を残します。それには、ストリーム版の filter⼿続き
(2.2.3節)を使います。
(define (stream-filter pred stream)

(cond ((stream-null? stream) the-empty-stream)
((pred (stream-car stream))
(cons-stream (stream-car stream)

(stream-filter
pred
(stream-cdr stream))))

(else (stream-filter pred (stream-cdr stream)))))

stream-filterは、ストリームの stream-car(ペアの car、つまり 10,000)を
テストします。これは素数ではないので、stream-filter は⼊⼒ストリーム
の stream-cdrを調べます。stream-cdrを呼び出すと、遅延化された stream-
enumerate-intervalの評価を強制します。評価すると、次のものが返ってき
ます。
(cons 10001

(delay (stream-enumerate-interval 10002 1000000)))

stream-filterは、今度はこのストリームの stream-car、つまり 10,001を⾒て、
これも素数ではないので、さらに stream-cdrを強制します。これを stream-
enumerate-intervalが素数 10,007を作り出すまで繰り返します。10,007が返
ってくると、stream-filterは定義通りに次のものを返します。
(cons-stream (stream-car stream)

(stream-filter pred (stream-cdr stream)))

57ここで⽰した数値は、実際には遅延オブジェクトの中にはありません。実際にあるの
は元の式で、環境の中で変数が適切な数値に束縛されています。例えば、10001と⽰した
場所には実際には (+ low 1)があり、その lowが 10,000に束縛されています。

この場合は、以下のようになります。
(cons 10007

(delay (stream-filter
prime?
(cons 10008

(delay (stream-enumerate-interval
10009
1000000))))))

この結果が、元の式の stream-cdr に渡されることになります。これによっ
て、遅延化された stream-filterが強制され、stream-filterは次の素数であ
る 10,009が⾒つかるまで遅延化された stream-enumerate-intervalの強制を
続けます。最終的に、元の式の stream-carに渡される結果は次のようになり
ます。
(cons 10009

(delay (stream-filter
prime?
(cons 10010

(delay (stream-enumerate-interval
10011
1000000))))))

stream-carは 10,009を返し、これで計算が完了します。⼆つ⽬の素数を⾒つ
けるために必要なだけの整数のみがテストされ、素数フィルタに与えるのに必
要な分だけの区間だけが列挙されています。
⼀般的に、遅延評価は “要求駆動”プログラミングと考えることができま

す。要求駆動プログラミングでは、ストリーム処理の各段階は、次の段階を満
⾜させるのに⼗分なだけしか動かされません。ここまででやったことは、計算
の中で起こることの実際の順番と、⼿続きの⾒かけ上の構造とを切り離すとい
うことです。⼿続きを書くときには、ストリームが “すべて同時に”存在するか
のように考えて書くのですが、実際には、計算は伝統的なプログラミングスタ
イルの場合と同じように漸進的に実⾏されることになります。

delayと forceの実装
delayと forceはミステリアスな演算に⾒えるかもしれませんが、実際のとこ
ろ、実装はとても単純です。delayは、後で要求されたときになって評価する

ように、式をパッケージする必要があります。これは、単純に式を⼿続きの本
体として扱うことによって実現できます。delayは、次のようなシンタックス
シュガーによって
(delay ⟨exp⟩)

以下のものを表すような特殊形式として実装できます。
(lambda () ⟨exp⟩)

forceは、delayによって⽣成された (引数なしの)⼿続きを呼び出すだけです。
ですので、forceは次のような⼿続きとして実装できます。
(define (force delayed-object) (delayed-object))

delayと forceが仕様通りに動くようにするだけなら、この実装で⼗分です。
しかし、重要な最適化をひとつ取り⼊れる余地があります。多くの応⽤では、
同じ遅延オブジェクトを何回も強制するはめになります。これは、ストリーム
を使った再帰プログラムで、深刻な⾮効率性の原因となることがあります (練
習問題 3.57参照)。これを解決するには、遅延オブジェクトが最初に強制され
る際に計算した値を格納するように遅延オブジェクトを構築します。⼆回⽬以
降の強制では、計算を繰り返すことはしないで、格納した値をそのまま返しま
す。つまり、練習問題 3.27で説明したものと同じような、特別⽬的のメモ化⼿
続きとして delayを実装するということです。これを実現する⽅法のひとつと
して、(引数なしの) ⼿続きをひとつ引数として取り、その⼿続きのメモ化バー
ジョンを返す⼿続きを使うというものがあります。以下にその⼿続きを⽰しま
す。メモ化⼿続きが⼀回⽬に実⾏されると、計算結果を格納します。⼆回⽬以
降の評価では、その結果をそのまま返します。
(define (memo-proc proc)

(let ((already-run? false) (result false))
(lambda ()

(if (not already-run?)
(begin (set! result (proc))

(set! already-run? true)
result)

result))))

そして、(delay <exp>)が以下と等価になるように delayを定義します。
(memo-proc (lambda () ⟨exp⟩))

forceは以前の定義そのままです。58

練習問題 3.50:以下の定義を完成させよ。これは、2.2.1節Footnote
12の map と同じように、複数の引数を取る⼿続きを使えるよう
stream-mapを⼀般化するものである。
(define (stream-map proc . argstreams)

(if (⟨??⟩ (car argstreams))
the-empty-stream
(⟨??⟩
(apply proc (map ⟨??⟩ argstreams))
(apply stream-map

(cons proc (map ⟨??⟩ argstreams))))))

練習問題 3.51: 遅延評価についてより詳しく観察するために、引
数を表⽰してからそのまま返すだけの以下の⼿続きを使うことに
する。
(define (show x)

(display-line x)
x)

以下のような⼀連の式を評価させると、インタプリタはどのよう
な表⽰をするだろうか。
59

58ストリームの実装は、この節で説明したもの以外にもいろいろ考えられます。スト
リームを実⽤的なものにする鍵となっている遅延評価は、Algol 60では 名前呼び (call-
by-name)という引数渡し⼿法に元々備わっていました。このメカニズムを使ってストリ
ームを実装するという考え⽅は、Landin (1965)で初めて記述されたものです。ストリー
ムの遅延評価は、Friedman and Wise (1976)によって Lispに導⼊されました。この実
装では、consは常に引数の評価を遅延させるようになっていて、そのためリストは⾃動
的にストリームとしてふるまうようになっていました。メモ化による最適化は、必要呼
び (call-by-need)としても知られています。Alogolコミュニティであれば、ここでの元
々の遅延オブジェクトは名前呼びサンク (call-by-name thunk)と呼び、最適化されたバ
ージョンは必要呼びサンク (call-by-need thunk)と呼ぶところです。

59練習問題 3.51や練習問題 3.52のような課題は、delayの動作についてちゃんと理解
しているかどうかを確認できるという意味で、価値のあるものです。⼀⽅、遅延評価と
表⽰を混在させると、頭が混乱しやすくなります。代⼊まで⼊ってくると、ますますひ
どくなります。コンピュータ⾔語の授業では、この節に出てきたような試験問題で講師

354



(define x
(stream-map show

(stream-enumerate-interval 0 10)))
(stream-ref x 5)
(stream-ref x 7)

練習問題 3.52: 以下の⼀連の式について考える。
(define sum 0)
(define (accum x) (set! sum (+ x sum)) sum)
(define seq

(stream-map accum
(stream-enumerate-interval 1 20)))

(define y (stream-filter even? seq))
(define z

(stream-filter (lambda (x) (= (remainder x 5) 0))
seq))

(stream-ref y 7)
(display-stream z)

上記の式をそれぞれ評価した後、sumの値はいくつになるだろう
か。式 stream-refと display-streamを評価したとき、どのよう
な応答が表⽰されるだろうか。(delay ⟨exp⟩) を単純に (lambda
() ⟨exp⟩)として実装し、memo-procによって提供される最適化を
使⽤しない場合、これらの応答は異なるものになるだろうか。説
明せよ。

3.5.2 無限ストリーム
ここまで、実際にはアクセスする必要のある分しかストリームの計算をしてい
ないのに、ストリームを完全な実体として扱っているという幻想を維持する⽅
が学⽣を苦しめるということが、伝統的によく⾏われてきています。⾔うまでもないこ
とですが、そのような難しいところに依存するようなプログラムを書くというのは、醜
いプログラミングスタイルです。ストリーム処理の⼒の⼀部は、プログラムの中で実際
に出来事が起こる順番を考えなくていいというところから来ています。残念ながら、代
⼊があると、このような考え⽅はできません。代⼊を使うときには、時間と変化につい
て考えることが避けられないからです。

法について⾒てきました。このテクニックを使うと、列が⾮常に⻑いものでも、
効率的にストリームとして表現できます。さらに特筆すべきこととして、スト
リームを使うと無限に⻑い列を表すこともできます。例えば、以下に⽰す正の
整数のストリームの定義について考えてみましょう。
(define (integers-starting-from n)

(cons-stream n (integers-starting-from (+ n 1))))
(define integers (integers-starting-from 1))

これは筋が通っています。integersは、carが 1で、cdrが “2から始まる整
数を⽣成する約束”というペアになります。これは無限に⻑いストリームです
が、しかしある時間のうちに調べられるのはそのその有限な⼀部だけです。そ
のため、プログラムからは無限ストリーム全体が存在するわけではないという
ことはわかりません。

integersを使って、別の無限ストリームを定義できます。例えば、7で割
り切れない整数のストリームは以下のようになります。
(define (divisible? x y) (= (remainder x y) 0))
(define no-sevens

(stream-filter (lambda (x) (not (divisible? x 7)))
integers))

こうすると、このストリームの要素にアクセスするだけで、7で割り切れない
整数を探すことができます。
(stream-ref no-sevens 100)
117

integersと同じようなやり⽅で、フィボナッチ数の無限ストリームを定義す
ることもできます。
(define (fibgen a b) (cons-stream a (fibgen b (+ a b))))
(define fibs (fibgen 0 1))

fibsは、carが 0で、cdrが “(fibgen 1 1)を評価する約束”というペアです。
この遅延化した (fibgen 1 1)を評価すると、carが 1で、cdrが “(fibgen 1
2)を評価する約束”というペアを⽣成し . . .という繰り返しになります。
もっと刺激的な無限ストリームについても⾒てみましょう。no-sevensの

例を⼀般化すると、エラトステネスのふるい (sieve of Eratosthenes)という⼿

法によって素数の無限ストリームを構築できます。60 まず、最初の素数である
2から始まる整数列を考えます。残りの素数を得るには、まず整数列の残りか
ら 2の倍数をフィルタリングします。こうすると、次の素数である 3から始ま
るストリームができます。次に、このストリームの残りから 3の倍数をフィル
タリングします。こうすると、次の素数である 5から始まるストリームができ
ます。これをずっと続けていきます。これは、⾔い換えると次のようなふるい
のプロセスで素数列を作るということになります。あるストリーム Sについて、
Sの最初の要素を最初の要素とし、Sの残りから Sの最初の要素の倍数をフィ
ルタリングした結果をふるいにかけたものを残りの要素とするようなストリー
ムを作ることによって、Sにふるいをかけるということです。このプロセスは、
ストリーム演算によってすぐに記述できます。
(define (sieve stream)

(cons-stream
(stream-car stream)
(sieve (stream-filter

(lambda (x)
(not (divisible? x (stream-car stream))))

(stream-cdr stream)))))
(define primes (sieve (integers-starting-from 2)))

これで、質問をするだけで特定の素数を得ることができるようになります。
(stream-ref primes 50)
233

sieveによって作られる信号処理システムに注⽬すると⾯⽩いことがわか
ります。このシステムを “ヘンダーソン図”として図 3.31に⽰します。61 ⼊⼒

60エラトステネスは紀元前 3世紀のアレキサンドリア学派のギリシャ⼈哲学者で、地
球の外周を最初に正しく推測したことで有名です。彼は、夏⾄の⽇の正午の影を観察す
ることによってそれを計算しました。エラトステネスのふるいは⾮常に古いものですが、
最近まで巨⼤素数を⾒つける最も強⼒なツールであった特別⽬的ハードウェアによる
“ふるい”は、このエラトステネスのふるいを基礎としています。しかし、70年代からは、
これらの⼿法は1.2.6節で検討したような確率的な技術の発展によって取って代わられて
います。

61私たちは Peter Hendersonにちなんでこれらの図をヘンダーソン図と呼んでいます。
ストリーム処理について考える⽅法のひとつとして、この種の図を初めて⾒せてくれた
のが Hendersonだったからです。各実線は、転送される値のストリームを表しています。
carから consと filterに伸びる点線は、これがストリームではなく、単⼀の値である
ことを⽰しています。

filter:
not
divisible?

sieve

sieve

car

cdr
cons

図 3.31: 信号処理システムとして⾒た素数のふるい

ストリームは “unconser”に流し込まれ、そこでストリームの最初の要素がス
トリームの残りから切り離されます。最初の要素を使って整除性フィルタが構
築され、ストリームの残りがそのフィルタを通ります。このフィルタの出⼒は、
また別のふるい箱に流されます。そして、元の最初の要素が内部のふるいの出
⼒と consされ、それが出⼒ストリームとなります。つまり、無限なのはスト
リームだけではなく、信号処理器も無限だということです。ふるいは中にふる
いを持っているからです。

ストリームの暗黙定義
上記の integersと fibsのストリームは、明⽰的にストリーム要素をひとつず
つ計算する “⽣成”⼿続きを規定することによって定義されていました。スト
リームを規定する別のやり⽅として、遅延評価を利⽤して暗黙的にストリーム
を定義するというものがあります。例えば、次の式は onesというストリーム
を 1の無限ストリームとして定義します。
(define ones (cons-stream 1 ones))

これは再帰⼿続きの定義と同じような仕組みです。onesは、carが 1で、cdr
が onesを評価する約束というペアになります。cdrを評価すると、1と、ones
を評価する約束がまた得られます。この繰り返しです。
演算によってストリームを操作すると、もっと⾯⽩いことができます。例

えば、add-streamsという、⼆つの与えられたストリームの要素ごとの和を⽣
成する演算について考えます。62

62これは、練習問題 3.50で出てきた⼀般化されたバージョンの stream-mapを利⽤し

(define (add-streams s1 s2) (stream-map + s1 s2))

こうすると、整数列を次のように定義できます。
(define integers

(cons-stream 1 (add-streams ones integers)))

これは、⼀つ⽬の要素が 1で、残りが onesと integersの和であるようなもの
として、integersを定義するものです。つまり、integersの⼆つ⽬の要素は、
1 + integersの⼀つ⽬の要素で、2になります。integersの三つ⽬の要素は、
1 + integersの⼆つ⽬の要素で、3になります。この繰り返しです。この定義
がうまくいくのは、どの時点でも⼗分な integersストリームが⽣成されてい
るので、それを定義にフィードバックすることによって次の整数が⽣成できる
からです。
フィボナッチ数も同じスタイルで定義できます。

(define fibs
(cons-stream
0
(cons-stream 1 (add-streams (stream-cdr fibs) fibs))))

この定義の意味は、fibsは 0, 1から始まるストリームで、残りのストリームは
⾃分⾃⾝に fibsをひとつずらしたものを⾜すことによって⽣成できるという
ことです。

1 1 2 3 5 8 13 21 . . . = (stream-cdr fibs)
0 1 1 2 3 5 8 13 . . . = fibs

0 1 1 2 3 5 8 13 21 34 . . . = fibs

そのようなストリーム定義を定式化するのに便利な⼿続きとしては、ほかに
scale-stream というものもあります。これは、ストリームの各要素に与えら
れた定数をかけるものです。
(define (scale-stream stream factor)

(stream-map (lambda (x) (* x factor))
stream))

例えば、
(define double (cons-stream 1 (scale-stream double 2)))

ています。

は、1, 2, 4, 8, 16, 32, . . . という 2の冪乗のストリームを⽣成します。
素数ストリームの別の定義の仕⽅は、整数列から始めて、素数性テストに

よってフィルタリングするというものです。まず、最初の素数である 2が必要
となります。
(define primes

(cons-stream
2
(stream-filter prime? (integers-starting-from 3))))

この定義は、ある整数 nが素数かどうかをテストするのに、√n以下の (整数
ではなく)素数で割り切れるかどうかをチェックすることになるので、思った
ほど単純なものにはなりません。
(define (prime? n)

(define (iter ps)
(cond ((> (square (stream-car ps)) n) true)

((divisible? n (stream-car ps)) false)
(else (iter (stream-cdr ps)))))

(iter primes))

これは再帰定義です。primesは prime?述語を使って定義されているのですが、
prime?⾃⾝は primesストリームを使っているからです。この⼿続きが動くの
は、どの時点でも、次にチェックする数の素数性をテストするのに⼗分なだけ
primesストリームが⽣成されているからです。つまり、素数性をテストするす
べての nについて、nは素数でないか (その場合、nを割り切る素数がすでに⽣
成されています)、素数であるか (その場合、√nより⼤きい素数—nよりは⼩
さい—がすでに⽣成されています) のどちらかです。63

練習問題 3.53: プログラムを実⾏することなしに、以下によって
定義されるストリームの要素を述べよ。

63この最後の点は⾮常に難しいところです。これは、pn+1 ≤ p2n という事実によりま
す (ここで、pk は k番⽬の素数を表しています)。このような推定は⽴証がとても難しい
ものです。素数が無限に存在するということに対する古代のユークリッドによる証明は
pn+1 ≤ p1p2 . . . pn + 1ということを⽰していますが、これよりずっとよい結果が証明
されたのは 1851年になってのことでした。この年、ロシア⼈の数学者 P. L. Chebyshev
が、すべての nに対し、pn+1 ≤ 2pn であることを証明しました。この結果は元々 1845
年に推測されていたもので、ベルトランの仮説 (Bertrand’s hypothesis)として知られて
います。証明はHardy and Wright 1960の 22.3節にあります。

(define s (cons-stream 1 (add-streams s s)))

練習問題 3.54: add-streamsと同じような仕組みで⼆つの⼊⼒ス
トリームの要素ごとの積を⽣成する⼿続き mul-streamsを定義せ
よ。これと integersストリームを組み合わせて、(0から数えて)n
番⽬の要素が n + 1の階乗となるような以下のストリームの定義
を完成させよ。
(define factorials

(cons-stream 1 (mul-streams ⟨??⟩ ⟨??⟩)))

練習問題 3.55:ストリーム Sを引数として取り、要素が S0, S0+S1,
S0 + S1 + S2, . . . であるストリームを返す⼿続き partial-sumsを
定義せよ。例えば、(partial-sums integers)は 1, 3, 6, 10, 15,
. . .というストリームになる。
練習問題 3.56: R. Hammingが提起した有名な問題に、素因数が
2, 3, 5のみからなる正の整数を昇順に重複なく列挙するというも
のがある。すぐ思いつく⽅法は、整数をひとつずつ⾒ていって、2,
3, 5以外の素因数がないかどうか調べるというものだ。しかし、こ
れは⾮常に⾮効率的である。これは、整数が⼤きくなるにつれて、
要求を満たすものがどんどん少なくなっていくことによる。別の
⽅法として、求める数値ストリームを Sとして、以下の事実に注
⽬してみる。

• Sは 1から始まる。
• (scale-stream S 2)の要素もまた、Sの要素である。
• 同じことが、(scale-stream S 3)と (scale-stream S 5)に
ついても⾔える。

• これらが Sの要素のすべてである。
つまり、これらの供給源から来る要素を組み合わせればよいとい
うことになる。そのために、⼆つの順序つきストリームを、重複を
除きつつ、ひとつの順序つきストリームにまとめたものを返す⼿
続き mergeを定義する。
(define (merge s1 s2)

(cond ((stream-null? s1) s2)

((stream-null? s2) s1)
(else
(let ((s1car (stream-car s1))

(s2car (stream-car s2)))
(cond ((< s1car s2car)

(cons-stream
s1car
(merge (stream-cdr s1) s2)))

((> s1car s2car)
(cons-stream
s2car
(merge s1 (stream-cdr s2))))

(else
(cons-stream
s1car
(merge (stream-cdr s1)

(stream-cdr s2)))))))))

こうすると、求めるストリームは mergeを使って次のように構築
できる。
(define S (cons-stream 1 (merge ⟨??⟩ ⟨??⟩)))

⟨?? ⟩となっている箇所に⽋けている式を埋めよ。
練習問題 3.57: add-streams⼿続きによる fibsの定義を使って n

番⽬のフィボナッチ数を計算する際に、加算は何回実⾏されるだ
ろうか。(delay <exp>)の実装を単純に (lambda () <exp>)とし
て、3.5.1節で述べた memo-proc⼿続きによって提供される最適化
を使わない場合、加算回数が指数的に増加することを⽰せ。64

練習問題 3.58: 以下の⼿続きによって計算されるストリームを解
釈せよ。

64この練習問題は、必要呼びが練習問題 3.27で述べた通常のメモ化と密接に関係して
いることを⽰しています。練習問題 3.27では、局所テーブルを明⽰的に構築するために
代⼊を使いましたが、ここでの必要呼びによるストリームの最適化では、ストリームの
以前に強制された部分の値を格納することによって、事実上、そのようなテーブルを⾃
動的に構築していることになります。

(define (expand num den radix)
(cons-stream
(quotient (* num radix) den)
(expand (remainder (* num radix) den) den radix)))

(quotientは、⼆つの整数に対して、整数の割り算の答えを返す基
本要素である。) (expand 1 7 10)はどのような要素列を⽣成する
だろうか。また、(expand 3 8 10)ではどうか。
練習問題 3.59: 2.5.3節では、多項式を項のリストとして表現する
多項式数値演算システムの実装⽅法を学んだ。似たような⽅法で、
以下のような冪級数 (power series)を、

ex = 1 + x+
x2

2
+

x3

3 · 2 +
x4

4 · 3 · 2 + . . . ,

cosx = 1− x2

2
+

x4

4 · 3 · 2 − . . . ,

sinx = x− x3

3 · 2 +
x5

5 · 4 · 3 · 2 − . . .

無限ストリームとして表現したものを扱うことができる。級数
a0 + a1x+ a2x

2 + a3x
3 + . . .は、係数 a0, a1, a2, a3, . . .を要素と

して持つストリームとして表すことにする。
a. 級数 a0 + a1x+ a2x

2 + a3x
3 + . . .の積分は次の級数になる。

c+ a0x+
1

2
a1x

2 +
1

3
a2x

3 +
1

4
a3x

4 + . . . ,

ここで、c は任意の定数である。羃級数を表すストリーム
a0, a1, a2, . . .を⼊⼒として取り、その級数の積分の係数 (定
数項を除く) のストリーム a0, 1

2
a1, 1

3
a2, . . . を返す⼿続き

integrate-series を定義せよ (結果は定数項を持たないた
め、羃級数を表していない。integrate-seriesを使う際に
は、適切な定数に consすることにする)。

b. 関数 x 7→ exは、⾃分⾃⾝の導関数となっている。つまり、ex
と、ex の積分が、定数項 e0 = 1を除いて同じ級数になると
いうことになる。したがって、exの級数は次のように⽣成で
きる。

(define exp-series
(cons-stream 1 (integrate-series exp-series)))

sinの導関数が cosで、cosの導関数が負の sinであるという
ことに基づき、sinと cosの級数を⽣成する⽅法を⽰せ。
(define cosine-series (cons-stream 1 ⟨??⟩))
(define sine-series (cons-stream 0 ⟨??⟩))

練習問題 3.60: 練習問題 3.59のように係数のストリームとして表
現された羃級数では、級数の加算は add-streamsにより実装され
る。級数を乗算する以下の⼿続きの定義を完成させよ。
(define (mul-series s1 s2)

(cons-stream ⟨??⟩ (add-streams ⟨??⟩ ⟨??⟩)))

練習問題 3.59の級数を使って sin2 x + cos2 x = 1となることを確
認することで、⼿続きをテストできる。
練習問題 3.61: S を、定数項が 1の羃級数 (練習問題 3.59)とする。
羃級数 1/S、つまり SX = 1となるような級数X を求めたいとす
る。SR を S の定数項の後の部分として、S = 1 + SR と書く。す
ると、X は以下のように解ける。

S ·X = 1,

(1 + SR) ·X = 1,

X + SR ·X = 1,

X = 1− SR ·X.

⾔い換えると、X は定数項が 1 で、⾼次の項が −SR · X として
与えられる羃級数ということになる。この考え⽅を使って、定数
項 1を持つ羃級数 S に対する 1/S を求める⼿続き invert-unit-
seriesを書け。練習問題 3.60の mul-seriesを使う必要がある。
練習問題 3.62: 練習問題 3.60と練習問題 3.61の結果を使って、2
つの羃級数を割る⼿続き div-seriesを定義せよ。div-seriesは
任意の 2つの級数について動作するようにする (もし分⺟の定数項
がゼロなら div-seriesはエラーを出すようにする)。div-series
と練習問題 3.59の結果と組み合わせて tanの羃級数を⽣成する⽅
法を⽰せ。

3.5.3 ストリームパラダイムの活⽤
遅延評価つきのストリームは強⼒なモデル化の道具として使うことができ、局
所状態や代⼊が持つ利点の多くを提供します。そのうえ、プログラミング⾔語
に代⼊を持ち込むときに⽣じる理論的に⾯倒なところをいくらか避けることが
できます。
ストリームというアプローチは、状態変数への代⼊を中⼼に組み⽴てるシ

ステムとは違うモジュール境界を持つシステムを構築できるという意味で、学
ぶところが多くあります。例えば、個別の瞬間における状態変数の値ではなく、
時系列 (または信号)全体を興味の中⼼に据えることができるようになります。
このことで、異なる瞬間の状態の構成要素を組み合わせたり⽐較したりするこ
とが便利になります。

ストリームプロセスとしての反復の定式化
1.2.1節では、状態変数を更新することによる反復プロセスを導⼊しました。今
では、状態を表現するのに、変数を更新していくのではなく、“時間のない”値
のストリームを使うことができます。この視点から1.1.7節の平⽅根⼿続きを
考え直してみましょう。そこでの考え⽅は、推測値を改善する⼿続きを何度も
適⽤することによって、xの平⽅根の推測値の列を⽣成するというものでした。
(define (sqrt-improve guess x)

(average guess (/ x guess)))

元の sqrt⼿続きでは、状態変数の⼀連の値がこれらの推測値となっていまし
た。その代わりにここで考える⽅法は、推測値の無限ストリームを⽣成すると
いうものです。ストリームの最初の値は初期値の 1です。65

(define (sqrt-stream x)
(define guesses

(cons-stream
1.0
(stream-map (lambda (guess) (sqrt-improve guess x))

guesses)))
guesses)

65let を使って局所変数 guesses を束縛することはできません。guesses の値は
guesses⾃⾝に依存しているからです。練習問題 3.63では、なぜここで局所変数が必要
なのかを考えます。

(display-stream (sqrt-stream 2))
1.
1.5
1.4166666666666665
1.4142156862745097
1.4142135623746899
. . .

ストリームの項を⽣成していくことによって、推測値を改善していくことがで
きます。答えが⼗分よいものになるまで項を⽣成し続ける⼿続きを書くことも
できます (練習問題 3.64参照)。
このほかに、1.3.1節で⾒た交代級数に基づいて πの近似値を⽣成する反復

も同じように扱えます。
π

4
= 1− 1

3
+

1

5
− 1

7
+ . . . .

まず、級数の加数 (項ごとに符号が逆になる奇数の逆数)のストリームを⽣成し
ます。次に、(練習問題 3.55の partial-sums⼿続きを使って)対象項数が増え
ていく部分和のストリームを作ります。それから結果を 4倍します。
(define (pi-summands n)

(cons-stream (/ 1.0 n)
(stream-map - (pi-summands (+ n 2)))))

(define pi-stream
(scale-stream (partial-sums (pi-summands 1)) 4))

(display-stream pi-stream)
4.
2.666666666666667
3.466666666666667
2.8952380952380956
3.3396825396825403
2.9760461760461765
3.2837384837384844
3.017071817071818
. . .

これは、収束は⽐較的遅いのですが、改善し続ける πの近似値のストリームに
なっています。列の 8個の項で、πの値は 3.284から 3.017の範囲になります。
ここまでの範囲では、状態のストリームというアプローチは、状態変数を

更新するというやり⽅とそれほど違いがありません。しかし、ストリームを
使うと、⾯⽩いトリックを使う余地が出てきます。例えば、列加速 (sequence
accelerator)というものを使うと、ストリームを変形して、近似値の列を同じ値
により速く収束する新しい列に変換するということができます。
そのような加速のひとつとして、18世紀のスイスの数学者レオンハルト・

オイラー (Leonhard Euler)によるものがありますが、これは交代級数 (項ごと
に符号が逆になる列)の部分和の列に対してうまく働きます。オイラーの⼿法
では、Sn を元の和の列の n番⽬の項とすると、加速した列は次のような項を
持つことになります。

Sn+1 −
(Sn+1 − Sn)

2

Sn−1 − 2Sn + Sn+1
.

したがって、元の列が値のストリームとして表されているなら、変換した列は
以下によって与えられます。
(define (euler-transform s)

(let ((s0 (stream-ref s 0)) ; Sn−1

(s1 (stream-ref s 1)) ; Sn

(s2 (stream-ref s 2))) ; Sn+1

(cons-stream (- s2 (/ (square (- s2 s1))
(+ s0 (* -2 s1) s2)))

(euler-transform (stream-cdr s)))))

これで、πの近似値の列を使ってオイラーの加速を実際にやってみることがで
きます。
(display-stream (euler-transform pi-stream))
3.166666666666667
3.1333333333333337
3.1452380952380956
3.13968253968254
3.1427128427128435
3.1408813408813416
3.142071817071818
3.1412548236077655

. . .

さらに都合のよいことに、加速した列はさらに加速でき、さらに再帰的にそれ
を加速し . . . と続けていくこともできます。つまり、ストリームのストリーム
(この構造については、ここでは タブロー (tableau)と呼ぶことにします)を作
ることになります。タブローの中では、それぞれのストリームはひとつ前のス
トリームを変換したものになっています。
(define (make-tableau transform s)

(cons-stream s (make-tableau transform (transform s))))

タブローは以下のような形をしています。
s00 s01 s02 s03 s04 . . .

s10 s11 s12 s13 . . .

s20 s21 s22 . . .

. . .

最後に、タブローの各⾏の最初の項を取って列を作ります。
(define (accelerated-sequence transform s)

(stream-map stream-car (make-tableau transform s)))

このタイプの “超加速”を πの列で実際にやってみます。
(display-stream
(accelerated-sequence euler-transform pi-stream))

4.
3.166666666666667
3.142105263157895
3.141599357319005
3.1415927140337785
3.1415926539752927
3.1415926535911765
3.141592653589778
. . .

得られる結果は⽬覚ましいものです。8個の項を取ると、⼩数点以下 14桁まで
正しい πの値が求められます。元の πの列だけ使っていたら、同じぐらい正確
な答えを求めようと思うと、1013項ぐらいまで計算する (つまり、各項が 10−13

より⼩さくなるまで級数を展開する) 必要があるところです。

これらの加速テクニックは、ストリームを使わないでも実装できたかもし
れません。しかし、ストリームによる定式化は、とりわけエレガントで便利で
す。状態列全体をデータ構造として使うことができ、それを統⼀された演算の
集合によって操作できるからです。

練習問題 3.63: Louis Reasonerは、sqrt-stream⼿続きを書くの
に、局所変数 guessesを使わない以下のような素直な実装をなぜ
しないのかと尋ねた。
(define (sqrt-stream x)

(cons-stream 1.0 (stream-map
(lambda (guess)

(sqrt-improve guess x))
(sqrt-stream x))))

Alyssa P. Hackerは、このバージョンは冗⻑な計算を⾏うのでだ
いぶ⾮効率的だと答えた。Alyssa の答えについて説明せよ。も
し delayの実装が memo-proc (3.5.1節) による最適化を使わずに
(lambda () <exp>)だけを使っていたとすると、それでも⼆つの
バージョンに効率の違いはあるだろうか。
練習問題 3.64: ストリームと数値 (許容誤差)を引数として取る⼿
続き stream-limitを書け。この⼿続きは、ストリームの⼆つの連
続する要素の差の絶対値が許容誤差以下になるまでストリームを
調べていき、その⼆つの値のうち後のものを返す。これを使うと、
与えられた許容誤差まで平⽅根を求めることができる。
(define (sqrt x tolerance)

(stream-limit (sqrt-stream x) tolerance))

練習問題 3.65: 級数

ln 2 = 1− 1

2
+

1

3
− 1

4
+ . . .

を使い、πについて上で⾏ったのと同様に、2の⾃然対数について
も近似値の列を三つ計算せよ。これらの列はどのくらい速く収束
するか。

ペアの無限ストリーム
2.2.3節では、列というパラダイムによって、伝統的な⼊れ⼦ループをペアの列
に対して定義されるプロセスとして扱う⽅法について⾒てきました。このテク
ニックを無限ストリームに⼀般化して、無限の集合について “ループ”しない
といけないためにループとして簡単に表現できないようなプログラムを書くこ
とができます。
例えば2.2.3 節の prime-sum-pairs ⼿続きを⼀般化して、i ≤ j の整数で

i + j が素数である整数ペア (i, j) すべてのストリームを作りたいとします。
int-pairsを i ≤ j の整数ペア (i, j)すべての列とすると、必要なストリームは
単純に次のようになります。66

(stream-filter
(lambda (pair) (prime? (+ (car pair) (cadr pair))))
int-pairs)

すると、問題は int-pairsストリームを⽣成することになります。より⼀般的
には、⼆つのストリーム S = (Si)と T = (Tj)があるとして、無限の⻑⽅形の
配列を想像してみましょう。

(S0, T0) (S0, T1) (S0, T2) . . .

(S1, T0) (S1, T1) (S1, T2) . . .

(S2, T0) (S2, T1) (S2, T2) . . .

. . .

⽣成したいのは、対⾓線上またはその上にあるペア、つまり以下に⽰すような
ペアすべてを含むストリームです。

(S0, T0) (S0, T1) (S0, T2) . . .

(S1, T1) (S1, T2) . . .

(S2, T2) . . .

. . .

(S と T を両⽅とも整数のストリームとすると、これが求めるストリーム int-
pairs になります。)
⼀般的なペアのストリームを (pairs S T)と呼び、それを三つの部分から

なるものとして考えます。ペア (S0, T0)、⼀⾏⽬の残りのペア、それ以外のペ
662.2.3節と同じように、ここでは整数ペアを Lispのペアとしてではなくリストとして
表現します。

アです。67

(S0, T0) (S0, T1) (S0, T2) . . .

(S1, T1) (S1, T2) . . .

(S2, T2) . . .

. . .

この分解の三つ⽬の部分 (⼀⾏⽬以外にあるペア) は、(stream-cdr S) と
(stream-cdr T)によって (再帰的に)作られるペアです。また、⼆番⽬の部分
(⼀⾏⽬の残り)は以下のように表現されます。
(stream-map (lambda (x) (list (stream-car s) x))

(stream-cdr t))

よって、求めるペアのストリームは次のように作ることができます。
(define (pairs s t)

(cons-stream
(list (stream-car s) (stream-car t))
(⟨何らかの⽅法で組み合わせる⟩

(stream-map (lambda (x) (list (stream-car s) x))
(stream-cdr t))

(pairs (stream-cdr s) (stream-cdr t)))))

⼿続きを完成させるには、⼆つの内部ストリームを組み合わせる何らかの⽅法
を選ぶ必要があります。考え⽅のひとつとして、2.2.1節の append⼿続きのス
トリーム版を使うというものがあります。
(define (stream-append s1 s2)

(if (stream-null? s1)
s2
(cons-stream (stream-car s1)

(stream-append (stream-cdr s1) s2))))

しかし、これは無限ストリームには向いていません。⼆つ⽬のストリームを組
み込む前に、⼀つ⽬のストリームから要素をすべて取ることになるからです。
具体的には、以下のようにしてすべての正の整数のペアを⽣成しようとすると、
(pairs integers integers)

67このような分解⽅法を選ぶ理由となる洞察については練習問題 3.68を参照してくだ
さい。

結果のストリームは、⼀つ⽬の要素が 1に等しいすべてのペアをまず列挙しよ
うとします。そのため、ほかの値が⼀つ⽬の要素となるようなペアは永遠に作
られないことになってしまいます。
無限ストリームを扱うには、プログラムを⼗分⻑い時間実⾏したら、すべ

ての要素にいつかは到達することを保証するような組み合わせの順番を考え出
す必要があります。これを達成するよくできた⽅法として、以下の interleave
⼿続きを使うというものがあります。68

(define (interleave s1 s2)
(if (stream-null? s1)

s2
(cons-stream (stream-car s1)

(interleave s2 (stream-cdr s1)))))

interleaveは要素を⼆つのストリームから交互に取るので、最初のストリー
ムが無限であっても、⼆つ⽬のストリームの要素はすべて、混ぜ合わせたスト
リームにいつかは⼊ることになります。
よって、求めるペアのストリームは以下のように⽣成できます。

(define (pairs s t)
(cons-stream
(list (stream-car s) (stream-car t))
(interleave
(stream-map (lambda (x) (list (stream-car s) x))

(stream-cdr t))
(pairs (stream-cdr s) (stream-cdr t)))))

練習問題 3.66: ストリーム (pairs integers integers)について
詳しく観察せよ。ペアがストリームに配置される順番について、全
体的なコメントをせよ。例えば、ペア (1, 100)の前にはおよそどれ
だけの数のペアが出てくるだろうか。(99, 100)と (100, 100)の場
合はどうだろうか。(正確な数学的表現ができるならなおよい。し
かし、⾏き詰まったら定性的な答えでもよい。)

68組み合わせの順番について求められる性質は、正確には、⼀つ⽬のストリームの i
番⽬の要素と⼆つ⽬のストリームの j 番⽬の要素に対応するペアが出⼒ストリームの
f(i, j)番⽬に出現するという⼆引数の関数 f が存在するということです。interleaveを
使ってこれを達成するトリックは David Turnerが⽰したもので、彼はKRC⾔語でこれ
を使っています。(Turner 1981)

練習問題 3.67: pairs⼿続きを修正し、(pairs integers integers)
がすべての整数のペア (i, j)を (i ≤ j という条件なしで)⽣成する
ようにせよ。ヒント：追加のストリームを混ぜる必要がある。
練習問題 3.68: Louis Reasonerは、ペアのストリームを構築する
のに三つの部分に分けるのは必要以上に複雑なのではないかと考
えた。彼は、ペア (S0, T0)を最初の⾏のそれ以外のペアと分ける代
わりに、次のように最初の⾏全体を扱うことを提案した。
(define (pairs s t)

(interleave
(stream-map (lambda (x) (list (stream-car s) x))

t)
(pairs (stream-cdr s) (stream-cdr t))))

これはうまくいくだろうか。(pairs integers integers)を Louis
の pairsの定義を使って評価した場合にどうなるか考えよ。
練習問題 3.69: 三つの無限ストリーム S, T , U を取り、i ≤ j ≤ k

の三つ組 (Si, Tj , Uk) のストリームを⽣成する⼿続き triples を
書け。triplesを使って、すべての正の整数のピタゴラス数の三つ
組、つまり i ≤ j かつ i2 + j2 = k2 であるような三つ組 (i, j, k)の
ストリームを⽣成せよ。
練習問題 3.70: ペアが出てくる順番が、アドホックな混ぜ合わせ
処理の結果ではなく、何か役に⽴つ順番となっているようなスト
リームが作れれば便利だ。もし、ある整数のペアがほかのペアと
⽐べて “⼩さい” かどうかを判断する⽅法を定義できるなら、練
習問題 3.56の merge ⼿続きに似たテクニックを使うことができ
る。これを⾏う⽅法のひとつとして、“重み関数”W (i, j)を定義し、
W (i1, j1) < W (i2, j2) なら (i1, j1) は (i2, j2) より⼩さいと取り決
めるというものがある。mergeに似ているが、追加の引数 weight
⼿続きを取る merge-weightedを書け。weightはペアの重みを計
算する⼿続きで、マージ結果のストリームの中で要素が現れる順
番を決めるのに使われる。69 これを使って、pairsを weighted-
pairsという⼿続きに⼀般化せよ。weighted-pairsは、⼆つのス

69ペアの配列の中で⾏沿いに右に移動したり、列沿いに下に移動したりすると、ペアの
重みが増加するようになっていることが必要になる。

トリームに加えて重み関数を計算する⼿続きを引数に取り、重み
に従って並べられたペアのストリームを⽣成する。その⼿続きを
使って、以下のそれぞれのストリームを⽣成せよ。

a. 和 i+ jの順に並んだ、すべての正の整数のペア (i, j)i ≤ jの
ストリーム

b. iと j のどちらも 2, 3, 5で割り切れないすべての正の整数の
ペア (i, j)i ≤ j が 2i+ 3j + 5ij の順に並んでいるストリーム

練習問題 3.71:⼆つの⽴⽅数の和で表す⽅法が複数ある数は、数学
者 Srinivasa Ramanujan(シュリニヴァーサ・ラマヌジャン) にち
なんでラマヌジャン数 (Ramanujan number)と呼ばれることがあ
る。70 ペアの順序つきストリームを使うと、エレガントな⽅法で
これらの数を計算できる。⼆つの⽴⽅数の和として表す⽅法が⼆
通りある数を⾒つけるには、i3 + j3 という和によって重みづけを
した整数のペア (i, j)のストリームを⽣成し (練習問題 3.70参照)、
それから同じ重みを持つ連続したペアをストリームから探すだけ
でよい。ラマヌジャン数を⽣成する⼿続きを書け。そのような最
初の数は 1,729である。次の五個を求めよ。
練習問題 3.72: 練習問題 3.71と似たようなやり⽅で、⼆つの平⽅
数の和として表す⽅法が三通りあるすべての数のストリームを⽣
成せよ (そのように書ける理由も⽰せ)。

信号としてのストリーム
ストリームの話を始めたとき、ストリームは信号処理システムの “信号”に対
するコンピュータ上での同等物であるという⾔い⽅をしました。実際、ストリ
ームを使うと、信号処理システムを⾮常に直接的なやり⽅でモデル化できます。
このとき、間隔を置いて連続した信号の値をストリームの連続した要素として

70以下は G. H. Hardy(ゴッドフレイ・ハロルド・ハーディ)によるラマヌジャン追悼
⽂ (Hardy 1921)からの引⽤です。“‘すべての⾃然数は彼の友達だった’と⾔ったのはリ
トルウッドだった (と思う)。彼が病気で Putneyにいたとき、⼀度彼を⾒舞いに⾏った
ときのことを覚えている。私が乗って⾏ったタクシーの番号は 1729だった。私は、この
数字はあまり⾯⽩くない、よくない兆しでなければいいが、と⾔った。すると彼は ‘いい
え、とても⾯⽩い数字ですよ。それは、⼆つの⽴⽅数の和として表す⽅法が⼆つある最
⼩の数です’と答えた。” 重みづけをしたペアを使ってラマヌジャン数を⽣成するトリッ
クは Charles Leisersonが⽰してくれたものです。

図 3.32: 信号処理システムとして⾒た integral⼿続き

表現します。例えば、ある⼊⼒ストリーム x = (xi)と初期値 C、⼩さい増分 dt

に対し、和
Si = C +

i∑
j=1

xj dt

を集積して、値 S = (Si) のストリームを返す積分器 (integrator) (加算器
(summer)とも)を実装できます。以下の integral⼿続きは、整数ストリーム
の “暗黙的なスタイル”の定義 (3.5.2節)に似ています。
(define (integral integrand initial-value dt)

(define int
(cons-stream initial-value

(add-streams (scale-stream integrand dt)
int)))

int)

図 3.32は、integral⼿続きに相当する信号処理システムの図です。⼊⼒スト
リームは dt倍され、加算器を通り、その出⼒がまた同じ加算器に戻されます。
intの定義にあった⾃⼰参照は、この図では加算器の出⼒を⼊⼒のひとつに接
続するフィードバックループとなっています。

練習問題 3.73: ストリームを使って時系列上の電流や電圧の値を
表すことによって電⼦回路をモデル化できる。例えば、抵抗値 R

の抵抗と静電容量 C のコンデンサが直列になっているRC 回路
(RC circuit)があるとする。注⼊電流 iに対する回路の電圧応答 v

は 図 3.33の式によって決まる。この構造は信号流れ図によって⽰
してある。

図 3.33: RC回路と対応する信号流れ図

この回路をモデル化した⼿続き RCを書け。RCは値 R, C, dtを⼊
⼒として取り、⼿続きを返す。返り値の⼿続きは、電流 iを表すス
トリームとコンデンサの初期電圧 v0 を引数として取り、出⼒とし
て電圧 vのストリームを⽣成する。例えば、RCを使って、(define
RC1 (RC 5 1 0.5))を評価することによって Rが 5[Ω]、C が 1[F]、
タイムステップが 0.5秒の RC回路をモデル化できる。これは、電
流の時系列を表すストリームとコンデンサの初期電圧を取り、電
圧の出⼒ストリームを⽣成する⼿続き RC1を定義する。
練習問題 3.74: Alyssa P. Hackerは物理センサから来る信号を処
理するシステムを設計している。彼⼥が作りたい重要な機能のひ
とつとして、⼊⼒信号のゼロ交差 (zero crossing)を記録する信号
というものがある。これは、⼊⼒信号が負から正に変わったとき
に結果の信号が +1となり、正から負に変わったときには −1とな
り、その他の場合には 0となるというものである (0という⼊⼒の
符号は正とする)。例えば、典型的な⼊⼒信号と、対応するゼロ交
差信号は以下のようになる。
. . . 1 2 1.5 1 0.5 -0.1 -2 -3 -2 -0.5 0.2 3 4 . . .

. . . 0 0 0 0 0 -1 0 0 0 0 1 0 0 . . .

376



Alyssa のシステムでは、センサからの信号はストリーム sense-
dataで表され、対応するゼロ交差のストリームはストリーム zero-
crossingsとする。Alyssaは、⼆つの値を引数として取り、値の符
号を⽐較し、0, 1, -1を適切に返す⼿続き sign-change-detector
を書いた。次に、ゼロ交差ストリームを以下のように構築した。
(define (make-zero-crossings input-stream last-value)

(cons-stream
(sign-change-detector
(stream-car input-stream)
last-value)

(make-zero-crossings
(stream-cdr input-stream)
(stream-car input-stream))))

(define zero-crossings
(make-zero-crossings sense-data 0))

そこに通りがかった Alyssaの上司 Eva Lu Atorは、このプログラ
ムは以下のプログラムとほぼ等価だと⾔った。このプログラムで
は、練習問題 3.50で出てきた、⼀般化したバージョンの stream-map
を使っている。
(define zero-crossings

(stream-map sign-change-detector
sense-data
⟨expression⟩))

⟨expression⟩ によって⽰された部分を埋め、プログラムを完成さ
せよ。
練習問題 3.75: 残念なことに、練習問題 3.74の Alyssaのゼロ交差
識別器は⼗分でないということがわかった。センサからノイズの
交じった信号が来ると、余計なゼロ交差を検出してしまうのだ。ハ
ードウェアのスペシャリスト Lem E. Tweakitは、ゼロ交差を検
出する前に信号を平滑化してノイズをフィルタリングすることを
Alyssaに提案した。Alyssaは彼のアドバイスに従い、センサデー
タの値と直前の値との平均を取ることによって構築した信号から
ゼロ交差を検出することにした。彼⼥はこの問題について助⼿の

Louis Reasonerに説明した。彼は、Alyssaのプログラムを次のよ
うに修正して、この考え⽅を実装しようとした。
(define (make-zero-crossings input-stream last-value)

(let ((avpt (/ (+ (stream-car input-stream)
last-value)

2)))
(cons-stream
(sign-change-detector avpt last-value)
(make-zero-crossings
(stream-cdr input-stream) avpt))))

これは Alyssaの計画を正しく実装していない。Louisが⼊れてし
まったバグを探し、プログラムの構造を変更せずに直せ (ヒント：
make-zero-crossingsの引数の数を増やす必要がある)。
練習問題 3.76: Eva Lu Atorは、練習問題 3.75での Louisのアプロ
ーチに批判的だ。彼が書いたプログラムは、平滑化演算とゼロ交
差検出が交じってしまっていて、モジュール化されていないとい
うのが彼⼥の意⾒だ。例えば、⼊⼒信号を調整するよりよい⽅法
を Alyssaが⾒つけていれば、検出器は変更する必要がなかったは
ずだ。Louisを⼿伝って、⼊⼒としてひとつのストリームを取り、
⼊⼒ストリームの連続する⼆つの要素の平均を要素とするストリ
ームを⽣成する⼿続き smoothを書け。次に、smoothをコンポー
ネントとして使い、ゼロ交差判定機をよりモジュール化されたス
タイルで実装せよ。

3.5.4 ストリームと遅延評価
前節の終わりのほうにあった integral⼿続きは、ストリームを使うことによ
ってフィードバックループを持つ信号処理システムをモデル化する⽅法を⽰し
ています。図 3.32に⽰した加算器のフィードバックループは、integralの内
部ストリーム intが⾃分⾃⾝によって定義されることによってモデル化されて
います。
(define int

(cons-stream
initial-value

y0

dy y
integralmap: f

図 3.34: ⽅程式 dy/dt = f(y)を解く “アナログ計算回路”

(add-streams (scale-stream integrand dt)
int)))

暗黙的定義のようなものを扱うインタプリタの能⼒は、cons-streamに組み込
まれている delayに依存しています。この delayがなかったとすると、イン
タプリタは cons-streamの⼆つの引数を評価してからでないと intを構築で
きないのですが、⼆つの引数を評価するには intがすでに定義されている必要
がある、ということになってしまいます。⼀般的に、delayはループを含む信
号処理システムをストリームを使ってモデル化するときに鍵となる部分です。
delayがなければ、出⼒を返す前に、信号処理コンポーネントに対する⼊⼒が
完全に評価されているようにモデルを定式化しなければいけません。この場合、
ループは禁⽌されることになります。
残念ながら、ループを含むシステムに対するストリームモデルを構築する

には、cons-streamの提供する “隠れた”delayだけでは⾜りず、それを超えて
delayを使うことが求められることがあります。例えば、図 3.34は、f を与え
られた関数として、微分⽅程式 dy/dt = f(y)を解く信号処理システムを⽰して
います。この図には、f を⼊⼒信号に適⽤するマップコンポーネントがありま
す。マップコンポーネントは、このような⽅程式を解くために実際に使われる
アナログ計算機回路と似たやり⽅で、積分器に対するフィードバックループに
接続されています。

yの初期値 y0 が与えられているとすると、モデル化に次のような⼿続きを
使うことが考えられます。
(define (solve f y0 dt)

(define y (integral dy y0 dt))
(define dy (stream-map f y))

y)

この⼿続きはうまくいきません。solveの⼀⾏⽬では integralを呼び出して
いますが、この呼び出しのためには⼊⼒ dyが定義されている必要があります。
しかし、dyは solveの⼆⾏⽬まで定義されません。
⼀⽅で、この定義の意図するところは理解できます。原則として、dy を

知らなくても、yのストリームの⽣成を始められるからです。実際、integral
等の多くのストリーム演算は、引数に関する部分的な情報しかない状況で答
えの⼀部を⽣成できるという意味で、cons-stream に似た性質を持っていま
す。integralでは、出⼒ストリームの最初の要素は initial-valueによって
与えられます。そのため、出⼒ストリームの最初の要素は被積分値 dyを評価
しなくても⽣成できます。yの最初の要素さえわかれば、solveの⼆⾏⽬にあ
る stream-mapが動作し、dyの最初の要素を⽣成します。これによって yの次
の要素が⽣成されます。以下、この繰り返しとなります。
この考え⽅を⽣かして、integralを定義し直し、被積分ストリームを 遅延

引数 (delayed argument)として受け⼊れるようにします。integralは、出⼒
ストリームの⼆番⽬以降の要素の⽣成が要求されて初めて、被積分値を force
し、評価されるようにします。
(define (integral delayed-integrand initial-value dt)

(define int
(cons-stream
initial-value
(let ((integrand (force delayed-integrand)))

(add-streams (scale-stream integrand dt) int))))
int)

これで、yの定義の中で dyの評価を遅延させることによって、solve⼿続きが
実装できます。71

(define (solve f y0 dt)
(define y (integral (delay dy) y0 dt))
(define dy (stream-map f y))
y)

71この⼿続きは、すべての Scheme実装で動くことが保証されているわけではありませ
ん。とはいえ、どの実装でも単純な変更で動作させることができます。問題は、Scheme
実装が内部定義を扱うやり⽅に微妙な違いがあるということにあります (4.1.6節参照)。

⼀般的に、integralを呼び出す側は、被積分引数を delayしなければいけな
いようになっています。solve⼿続きがちゃんと動作するということは、初期
条件 y(0) = 1として、微分⽅程式 dy/dt = yの解について y = 1での値を求め
ることで e ≈ 2.718の近似値を求めることによって⽰すことができます。
(stream-ref (solve (lambda (y) y)

1
0.001)

1000)
2.716924

練習問題 3.77:上で使った integral⼿続きは、3.5.2節に出てきた、
整数の無限ストリームの “暗黙的”定義に似ている。別のやり⽅と
して、integralの定義を、(同じく3.5.2節に出てきた)integers-
starting-fromに似たものにすることもできる。
(define (integral integrand initial-value dt)

(cons-stream
initial-value
(if (stream-null? integrand)

the-empty-stream
(integral (stream-cdr integrand)

(+ (* dt (stream-car integrand))
initial-value)

dt))))

ループを含むシステムについて使うと、この⼿続きは integralの
元のバージョンと同じ問題を持つことになる。⼿続きを修正し、遅
延された引数を integrandとして受け⼊れるようにして、上に⽰
した solve⼿続きで使えるようにせよ。

練習問題 3.78: 同次 2階線形微分⽅程式について調べるために信
号処理システムを設計するという問題について考えよ。

d2y

dt2
− ady

dt
− by = 0.

yをモデル化した出⼒ストリームは、ループを含む回路によって⽣
成される。これは、d2y/dt2 の値が yと dy/dtの値に依存し、また
これらの値は両⽅とも d2y/dt2 を積分することによって決まるか

ddy

y0

dy y

scale: b

integralintegral

scale: a

add

dy0

図 3.35: ⼆次線形微分⽅程式を解くための信号フロー図

らである。コード化したいのは、図 3.35に⽰すようなものだ。引
数として、定数 a, b, dtと、yの初期値 y0と dy/dtの初期値 dy0を
取り、y の⼀連の値のストリームを⽣成する⼿続き solve-2ndを
書け。
練習問題 3.79: 練習問題 3.78の solve-2nd⼿続きを⼀般化し、⼀
般的な⼆次微分⽅程式 d2y/dt2 = f(dy/dt, y)を解くのに使えるよ
うにせよ。
練習問題 3.80: 直列RLC回路 (series RLC circuit)とは、抵抗・
コンデンサ・コイルが 図 3.36のように直列に接続されているもの
をいう。ここで、R, L, C がそれぞれ抵抗・コイル・コンデンサで
あるとすると、これら三つのコンポーネントの電圧 (v)と電流 (i)
の関係は、以下の⽅程式に⽰すようなものである。

vR = iRR, vL = L
diL
dt

, iC = C
dvC
dt

.

また、回路の接続によって、次の関係が成⽴する。
iR = iL = −iC , vC = vL + vR .

R

+

図 3.36: 直列 RLC回路

これらの式を組み合わせると、回路の状態 (コンデンサの電圧 vC
とコイルの電流 iL によって表される)は、以下の⼆つの微分⽅程
式によって記述される。

dvC
dt

= − iL
C
,

diL
dt

=
1

L
vC −

R

L
iL .

この微分⽅程式のシステムを表す信号フロー図を図 3.37に⽰す。
引数として、回路のパラメータR, L, Cと、時間の増分 dtを取る⼿
続き RLCを書け。練習問題 3.73の RC⼿続きと同じように、RLCは
⼿続きを返す。返り値の⼿続きは、状態変数の初期値 vC0 と iL0 を
取り、vC と iL の状態ストリームのペアを (consを使って)⽣成す
る。RLCを使って、R = 1[Ω], C = 0.2[F], L = 1 henry, dt = 0.1[s]、
また初期値を iL0 = 0[A], vC0 = 10[V]としたときの連結 RLC回
路のふるまいをモデル化するストリームのペアを⽣成せよ。

正規順序評価
この節で⽰した例は、delayと forceを明⽰的に使うことによってプログラミ
ングを柔軟にできるということを⽰していますが、⼀部の例からは、このこと
によってプログラムがより複雑になってしまうということもわかります。例え
ば、新しい integral⼿続きを使うことによってループを含むシステムをモデ
ル化する⼒が得られるのですが、こうすると integralが遅延化された被積分
関数を引数として呼び出す必要があるということを覚えておかないといけない
ことになり、integralを使う⼿続きはすべてこのことを認識していなければ

383



diL

vC0

iL

vC

dvC

iL0

scale: 1/L

integral

図 3.37: 直列 RLC回路を解くための信号フロー図

なりません。ここでは事実上、通常の⼿続きと、遅延化された引数を取る⼿続
きという⼆つのクラスの⼿続きを作ったことになります。⼀般的に、別々のク
ラスの⼿続きを作ると、別々のクラスの⾼階⼿続きを同じように作らなければ
いけないことになってしまいます。72

72Pascalのような従来型の強い型付け⾔語で⾼階⼿続きを扱おうとすると⾯倒なこと
になるのですが、この件は、Lispにそれが少し反映されたものです。強い型付け⾔語で
は、引数や返り値の型を⼿続きごとにプログラマが指定しなければいけません。数値、論
理値、配列などです。その結果として、“与えられた⼿続き procを列のすべての要素に
マップする”といった抽象化を、stream-mapのようなひとつの⾼階⼿続きによって表現
するということができません。その代わりに、procが取るかもしれない引数と返り値の
データ型の組み合わせすべてに対して、別々のマップ⼿続きが必要になってしまいます。
⾼階⼿続きの存在を考えたとき、実⽤的な概念として “データ型”というものを維持しよ
うとすると、いろいろと難しい問題が起こります。⾔語ML(Gordon et al. 1979)は、こ

⼆つの別々のクラスの⼿続きを作るはめになることを避けるためのひとつ
の⽅法として、すべての⼿続きが遅延化された引数を取るようにするというも
のがあります。⼿続きの引数がすべて⾃動的に遅延化され、実際に必要になっ
たとき (例えば、基本演算で必要になったとき)になって初めて強制されるよう
な評価モデルを採⽤することも考えられます。こうすると、私たちの⾔語は正
規順序評価を使うものになります。正規順序評価については、1.1.5節で評価の
置換モデルを導⼊した際に説明しました。正規順序評価に変えることにすると、
遅延評価を統⼀的でエレガントなやり⽅で扱うことができるようになります。
もしストリーム処理だけについて考えるのであれば、この戦略を採⽤するのが
⾃然でしょう。4.2節では、評価器について学んだ後に、私たちの⾔語をこの
⽅式に転換するやり⽅を⾒ていきます。残念ながら、⼿続き呼び出しに遅延を
採⽤すると、代⼊・可変データ・⼊出⼒といった、イベントの順番に依存する
プログラムを設計する能⼒がめちゃくちゃになってしまいます。cons-stream
で delayをひとつ使うだけでも、練習問題 3.51や練習問題 3.52で例⽰したよ
うに、ひどい混乱が起こることがあります。知られている限りでは、可変性と
遅延評価はプログラミング⾔語の中でうまく同居させることができません。こ
れら⼆つを同時に扱う⽅法を⼯夫することは活発な研究領域となっています。

3.5.5 関数型プログラムのモジュール性とオブジェクトの
モジュール性

3.1.2節で⾒たように、代⼊を導⼊することの主な利点のひとつは、⼤きなシス
テムの状態の⼀部をカプセル化する、つまり “隠す”ことによって、システム
のモジュール性を⾼められるというところにあります。ストリームモデルは、
代⼊を使わずに同等なモジュール性を提供します。ひとつの実例として、3.1.2
節で詳しく調べたモンテカルロ法による πの推定を、ストリーム処理という視
点から再実装してみます。
モジュール性という問題の鍵となっていたのは、乱数を使うプログラムか

ら、乱数⽣成器の内部状態を隠したいということでした。そのときは、まず
rand-updateという⼿続きを作りました。この連続する値が乱数を供給するこ
とになり、これを使って乱数⽣成器を⽣成するということをしていました。
の問題に対処する⽅法のひとつを⽰しています。この⾔語の “多相データ型”は、データ
型同⼠の⾼階変換のためのテンプレートを含んでいます。そのうえ、MLのほとんどの⼿
続きで、データ型がプログラマによって明⽰的に宣⾔されることはありません。その代
わりに、MLには型推論 (type-inferencing)メカニズムがあり、環境の情報を使って新し
く定義された⼿続きのデータ型を推論します。

385



(define rand
(let ((x random-init))

(lambda ()
(set! x (rand-update x))
x)))

ストリームによって定式化する場合は、乱数⽣成器そのものは存在せず、rand-
update を連続して呼ぶことによって⽣成される乱数ストリームがあるだけ
です。
(define random-numbers

(cons-stream
random-init
(stream-map rand-update random-numbers)))

これによって、rand-numbers ストリームの連続したペアに対してチェザロ
(Cesàro)実験を⾏った結果のストリームを構築します。
(define cesaro-stream

(map-successive-pairs
(lambda (r1 r2) (= (gcd r1 r2) 1))
random-numbers))

(define (map-successive-pairs f s)
(cons-stream
(f (stream-car s) (stream-car (stream-cdr s)))
(map-successive-pairs f (stream-cdr (stream-cdr s)))))

cesaro-streamは、次に monte-carlo⼿続きに供給されます。この⼿続きは、
確率の推定値のストリームを⽣成します。この結果が、今度は πの推定値のス
トリームに変換されます。このプログラムのこのバージョンでは、試⾏を何回
⾏うかという引数は必要ありません。(より多くの試⾏を⾏うことによる)より
よい π の推定値は、piストリームを後ろのほうまで取得していくことによっ
て得られます。
(define (monte-carlo experiment-stream passed failed)

(define (next passed failed)
(cons-stream
(/ passed (+ passed failed))
(monte-carlo

(stream-cdr experiment-stream) passed failed)))
(if (stream-car experiment-stream)

(next (+ passed 1) failed)
(next passed (+ failed 1))))

(define pi
(stream-map
(lambda (p) (sqrt (/ 6 p)))
(monte-carlo cesaro-stream 0 0)))

このアプローチにはかなりのモジュール性があります。このアプローチでも、
任意の実験を扱える⼀般的な monte-carlo⼿続きが定式化できているからで
す。それなのに、代⼊や局所変数といったものは使われていません。

練習問題 3.81: 練習問題 3.6では、乱数⽣成器を⼀般化して、乱数
列をリセットできるようにすることで、再現可能な “ランダム”な
数の列を⽣成することについて検討した。同じ⽣成器をストリー
ムによって定式化せよ。この⽣成器は要求の⼊⼒ストリームに基
づいて動作し、望み通りの乱数ストリームを⽣成する。要求は、新
しい乱数を⽣成する generate、または列を指定された値にリセッ
トする resetである。解答には代⼊を使わないこと。
練習問題 3.82: 練習問題 3.5のモンテカルロ積分をストリームによ
って再度⾏え。ストリーム版 estimate-integralは試⾏回数の引
数を持たない。その代わりに、連続した試⾏に基づく推定値のス
トリームを⽣成する。

時間の関数型プログラミング的視点
ここで、この章の初めで提起したオブジェクトと状態の問題に戻って、これら
の問題を新しい観点から検討してみましょう。この章の初めでは代⼊と可変オ
ブジェクトを導⼊し、状態を持つシステムをモデル化するプログラムをモジュ
ール式に構築するメカニズムを提供しました。また、局所変数を持つ計算オブ
ジェクトを構築し、これらの変数を代⼊によって変更しました。そして、この
世界に存在するオブジェクトの経時的なふるまいを、対応する計算オブジェク
トの経時的なふるまいによってモデル化しました。
その後、ストリームを使うことによっても、局所状態を持つオブジェクトを

モデル化できるということを⾒てきました。オブジェクトの局所状態のような
変化していく量は、逐次的な状態の時間履歴を表現するストリームによってモ

デル化できます。本質的に、ストリームによって時間を明⽰的に表現すること
によって、シミュレートされた世界の中での時間と、評価の間に起こるイベン
トの順番とを切り離すということをしていることになります。実際に、delay
というものが存在するので、モデル中のシミュレートされた時間と、評価の間
に起こるイベントの順番との間に、ほとんど関係がないということもありえ
ます。
これら⼆つのモデル化アプローチの違いを⾒るために、銀⾏⼝座の残⾼を

管理する “引き出し処理器”の実装について再度考えてみましょう。3.1.3節で
は、引き出し処理器の簡単なバージョンを実装しました。
(define (make-simplified-withdraw balance)

(lambda (amount)
(set! balance (- balance amount))
balance))

make-simplified-withdraw を呼び出すと計算オブジェクトが⽣成されます。
各オブジェクトはそれぞれ局所状態変数 balanceを持っています。balanceは、
オブジェクトに対する呼び出しを⾏うことで減少していきます。オブジェクト
は amount(⾦額)を引数として取り、新しい残⾼を返します。銀⾏⼝座のユーザ
ーが、そのようなオブジェクトに⼊⼒列をタイプし、ディスプレイの画⾯に返
り値の列が表⽰されるのを⾒ているところを想像するとよいでしょう。
引き出し処理器をモデル化する別のやり⽅として、⼊⼒として残⾼と引き

出す⾦額のストリームを取り、⼀連の⼝座残⾼のストリームを⽣成する⼿続き
を使うというものがあります。
(define (stream-withdraw balance amount-stream)

(cons-stream
balance
(stream-withdraw (- balance (stream-car amount-stream))

(stream-cdr amount-stream))))

stream-withdrawは、出⼒が⼊⼒によって完全に決定される、明確に定義され
た数学関数を実装しています。しかし、ここで⼊⼒ amount-streamがユーザー
のタイプする⼀連の値のストリームで、結果の残⾼ストリームは画⾯に表⽰さ
れていると考えてみてください。そうすると、値をタイプし結果を⾒ているユ
ーザーの視点からは、このストリーム処理は make-simplified-withdrawによ
って作られたオブジェクトと同じふるまいを持っていることになります。しか
しストリーム版では、代⼊も局所状態もなく、このため3.1.3節で直⾯したよう
な理論的な困難もありません。それなのに、システムには状態があるのです！

これは実に注⽬すべきところです。stream-withdrawは、ふるまいの変化
しない、明確に定義された数学関数を実装しているのに、ここでのユーザーの
知覚は、変化する状態を持つシステムと対話しているときと同じなのです。こ
のパラドックスを解くひとつの⽅法は、状態を持つことをシステムに求めてい
るのはユーザーの時間的な存在だということを認識するということです。もし
ユーザーがシステムとの対話から⼀歩下がって、個々の取引ベースで考えるの
をやめて残⾼のストリームという考え⽅をするなら、システムは状態を持たな
いように⾒えるでしょう。73

複雑な処理のある⼀部から⾒ると、ほかの部分は時間とともに変化するよ
うに⾒えます。それらは時間によって変わる局所状態を持っています。私たち
の世界に存在するこの種の⾃然な分解⽅法 (これは、私たちが世界の⼀部とし
ての視点から世界を⾒るやり⽅と同じです)をモデル化したプログラムをコン
ピュータ中の構造を使って書こうと思うと、関数的でない計算オブジェクトを
作ることになります。オブジェクトは変化しなければならないからです。状態
は局所状態変数によってモデル化し、状態の変更はこれらの変数に対する代⼊
によってモデル化します。こうすることによって、計算実⾏の時間が私たちの
いるこの世界の時間をモデル化するようにします。その結果、コンピュータの
中に “オブジェクト”ができることになります。
オブジェクトを使ったモデル化は強⼒で直感的です。その⼤きな理由とな

っているのは、このやり⽅が私たちがこの世界と関わる際の認知にマッチして
いるということです。しかし、この章を通して何度も⾒てきたように、これ
らのモデルは、イベント順の制約や、複数プロセスの同期といった、厄介な
問題を引き起こします。関数型プログラミング⾔語 (functional programming
language)の発展は、これらの問題を避けられるという⾒込みによって促進さ
れてきました。関数型プログラミング⾔語は代⼊や変更可能なデータを提供し
ません。このような⾔語では、すべての⼿続きは引数に対する明確な数学関数
を実装することになり、挙動は変化しません。並⾏システムを扱うには、関数
型のアプローチは⾮常に魅⼒的です。74

その⼀⽅、詳しく⾒ると、時間に関する問題が関数型のモデルにも紛れ込
73物理でも同じように、動く粒⼦を観測しているとき、粒⼦の位置 (状態)は変化して
いるというように⾔います。しかし、時空の中での粒⼦の世界線という視点からは、変
化というものは存在しません。

74Fortranの開発者 John Backusが 1978年に acmチューリング賞を受賞したとき、彼
は関数型プログラミングを⼀躍有名にしました。彼の受賞スピーチ (Backus 1978)は、関
数型のアプローチを強く⽀持するものでした。関数型プログラミングについてのよい概
説としては、Henderson 1980とDarlington et al. 1982があります。

merge bank
accountPaul's requests

Peter's requests

図 3.38: 取引要求リクエストのストリーム⼆つをマージす
ることでモデル化した共同銀⾏⼝座

んでいることがわかります。問題が多いのは、インタラクティブなシステムを
設計しようとする場合です。中でも、独⽴した実体同⼠のやりとりをモデル化
するようなシステムで特に問題となります。例えば、共同⼝座を持てるような
銀⾏システムの実装について、もう⼀度考えてみましょう。代⼊とオブジェク
トを使う従来型のシステムでは、Peterと Paulが⼝座を共有しているという事
実を、3.1.3節で⾒たように、Peterと Paulが取引要求を同じ銀⾏⼝座オブジ
ェクトに送るようにすることによってモデル化します。ストリームという視点
からは、“オブジェクト”そのものはなく、銀⾏⼝座のモデル化は、取引要求の
ストリームに対して演算を⾏い応答のストリームを⽣成する処理という形にな
るということは、すでに述べました。ということは、Peterと Paulが共同⼝座
を持っているという事実は、Peterの取引要求ストリームと Paulの取引要求ス
トリームをマージし、その結果を銀⾏⼝座ストリーム処理に送るという形でモ
デル化できるのではないでしょうか。この様⼦を図 3.38に⽰します。
この定式化で問題になるのは、マージ (merge)という概念です。⼆つのス

トリームをマージするのに、Peterの要求をひとつ、Paulの要求をひとつ、と
いった具合に、単純に交互に取っていくというやり⽅ではうまくいきません。
Paulが⼝座にめったにアクセスしないとしましょう。⼆回⽬の取引をするのは
Paulが⼝座にアクセスするまで待てと Peterに⾔うというのは無理な話です。
マージをどのように実装するにせよ、実装する際には、何らかの形で Peterと
Paulが知覚する “実時間”によって制約されるように、⼆つの取引ストリーム
を織り交ぜなければなりません。これはつまり、Peterと Paulが会ったとする
と、ある⼀部の取引は会う前に処理されていて、残りの取引は会った後に処理
されるということについて、合意できるようにするということです。75 これは

75どんな⼆つのストリームについても、それらを織り交ぜる適切な順番は、⼀般的に複
数あるということに注意してください。このため、厳密に⾔うと、“マージ”は関数では
なく関係ということになります。答えは⼊⼒に対する決定的な関数ではないということ
です。並⾏性を扱ううえで、⾮決定性が本質となっているということはすでに述べまし

390



まさに、3.4.1節で扱うことになった制約と同じものです。そのときは、状態を
持つオブジェクトを並⾏処理するにあたって、イベントの “正しい”順番を保
証するために、明⽰的な同期を導⼊する必要があることがわかりました。つま
り、関数型スタイルをサポートしようとしていたところで、複数の主体からの
⼊⼒をマージする必要が出てくると、関数型スタイルによって排除しようとし
ていた問題と同じものがまた現れるということになります。
この章は、モデル化対象である実世界の認知にマッチした構造を持つ計算

モデルを構築することをゴールとするところから始めました。世界をモデル化
するのには、分離した、時間制約のある、相互作⽤する、状態を持つオブジェ
クトの集まりとして扱うというやり⽅があり、単⼀の、時間のない、状態のな
いまとまりというやり⽅もあります。どちらの⾒⽅にも強⼒な利点があります
が、どちらの⾒⽅も単独では完全に⼗分ではありません。⼤統⼀は未だ現れて
いません。76

た (Footnote 39)。マージという関係は、この同じ本質的な⾮決定性について、関数型と
いう視点から⽰したものです。4.3節では、⾮決定性をまた別の視点から⾒ていくことに
なります。

76オブジェクトモデルは、世界を独⽴した部分に分けることによって近似します。関
数型モデルは、オブジェクト境界によってモジュール化するということはしません。オ
ブジェクトモデルは、“オブジェクト”の共有されていない状態が、共有されている状態
よりもずっと⼤きい場合に有⽤です。オブジェクトという視点がうまくいかない場合の
ひとつの例として量⼦⼒学があります。量⼦⼒学では、物を個別の粒⼦として考えると、
パラドックスや混乱が⽣まれます。オブジェクトという視点と関数型の視点を統⼀する
ということは、プログラミングとはあまり関係がなく、根本的な認識論の問題と関係が
あるのかもしれません。

4
メタ⾔語抽象化

. . . 魔法は⾔葉に宿る—アブラカダブラ、開けゴマ、などなど—し
かし、あるお話の魔法の⾔葉は、次のお話では魔法を持たない。本
当の魔法は、どの⾔葉が、いつ、何に対して効くかということを理
解するというところにある。そのコツは、コツを学ぶことにある
んだ。
. . . そして、それらの⾔葉は、普通のアルファベットの⽂字ででき
ている。20かそこらの、ペンで描けるぐにゃぐにゃした線。これ
こそが鍵なんだ！それに⼿が届きさえすれば、宝物にもなる。ま
るで—まるで宝物の鍵そのものが宝物みたいなんだ！
—John Barth, Chimera

プログラム設計について学んだ際に、熟練プログラマが設計の複雑さをコント
ロールするやり⽅は、ほかの複雑なシステムを設計する⼈たちが使うのと同じ
⼀般的なテクニックだということを⾒てきました。基本要素を組み合わせて複
合オブジェクトを作り、複合オブジェクトを抽象化して⾼レベルの構築⽤ブロ
ックを作り、適切なマクロ的な⾒⽅でシステム構造を⾒渡すことによってモ
ジュール性を維持するのです。これらのテクニックを説明するにあたっては、

Lispを使ってきました。Lispは、プロセスを記述し、また計算データオブジェ
クトやプロセスを構築して実世界の複雑な現象をモデル化するための⾔語とし
て役割を果たしてきました。しかし、取り組む問題が複雑になっていくにつれ、
Lispが、というよりもどんな固定したプログラミング⾔語でも、必要なことを
やり遂げるのに⼗分ではないと感じるようになります。考えをより効率的に表
現しようとすると、絶えず新しい⾔語に⽬を向けることになります。新しい⾔
語を作るというのは、⼯学的設計の複雑さをコントロールする強⼒な戦略です。
複雑な問題に取り組む際に、問題を別のやり⽅で記述する (また、それによっ
て思考する)ことのできる新しい⾔語を採⽤することで、私たち⾃⾝の能⼒を
向上させるということはよくあります。⼿元の問題に特にうまく合った基本要
素、組み合わせ⼿段、抽象化⼿段を持つ⾔語を使うということです。1

プログラミングは多くの⾔語に恵まれています。⾔語には、特定のコンピ
ュータの機械語といった物理的⾔語もあります。物理的⾔語が扱うのは、スト
レージの個々のビットや基本機械語命令というレベルでのデータや制御の表現
です。機械語のプログラマは、与えられたハードウェアを使って、資源に制約
のある計算を効率的に実装するシステムやユーティリティを構築することに焦
点を当てています。⾼級⾔語は、機械語という基層の上に構築され、ビットの
集合としてのデータの表現や基本命令の列としてのプログラムの表現といった
ところを気にしないでいいように隠してくれます。これらの⾔語には、⼿続き
定義のような組み合わせと抽象化の⼿段があります。これらの⼿段は⼤規模な
システムの構成に適しています。
メタ⾔語抽象化 (Metalinguistic abstraction)—新しい⾔語を構築すること—

1この考え⽅は、⼯学全体に広く⾏き渡っているものです。例えば、電気⼯学では、
回路の記述にさまざまな⾔語を使います。その中の⼆つとして、電気的 ネットワーク
(network)の⾔語と、電気的システム (system)の⾔語があります。ネットワーク⾔語は、
離散的電⼦素⼦によって物理的に装置をモデル化することに焦点を当てています。ネッ
トワーク⾔語の基本オブジェクトは、抵抗・コンデンサ・コイル・トランジスタなどの
基本的な電気部品で、それらは電流と電圧という物理的変数を特性として持ちます。ネ
ットワーク⾔語で回路を記述する際には、エンジニアが注⽬しているのは設計の物理的
特性です。それに対して、システム⾔語の基本オブジェクトは、フィルタやアンプといっ
た信号処理モジュールです。ここで関わってくるのはこれらのモジュールの機能的ふる
まいのみで、信号を操作するにあたってはモジュールの電圧や電流といった物理的実体
は気にしません。信号処理システムの要素が電気的ネットワークによって構築されてい
るという意味で、システム⾔語はネットワーク⾔語の上に成り⽴っているものです。し
かし、やりたいことは、与えられた応⽤問題を解くために電⼦装置を⼤規模に構成する
ということです。部品の物理的実現可能性は前提となっています。このような階層化さ
れた⾔語の集合は、2.2.4節の図形⾔語で⽰したような階層化設計テクニックのまた別の
例となっています。

これは、⼯学設計のすべての分野で重要な役割を果たしています。コンピュー
タプログラミングでは特に重要です。プログラミングでは、新しい⾔語を策定
することができるうえに、評価器を構築することによってそれらの⾔語を実装
することもできるからです。あるプログラミング⾔語の 評価器 (evaluator)(イ
ンタプリタ (interpreter)とも)とは、その⾔語の式に適⽤されたとき、その式
を評価するのに必要なアクションを実⾏するような⼿続きを指します。
次のことは、プログラミングについての最も根本的な考え⽅といっても過

⾔ではありません。
プログラミング⾔語の式の意味を決める評価器もまた、ただのプ
ログラムにすぎない。

この点を理解すると、プログラマとしての私たち⾃⾝のイメージが変わります。
他⼈の設計した⾔語のただのユーザーではなく、⾔語の設計者として⾃分⾃⾝
を⾒るようになります。
実際、ほとんどのプログラムは何らかの⾔語の評価器であると考えること

ができます。例えば、2.5.3節の多項式操作システムは、多項式の数値演算の規
則を具体化し、リスト構造のデータに対する演算という形で実装したものです。
このシステムを拡張して、多項式を読み込み、表⽰する⼿続きを追加したら、
記号数学の問題を扱う特定⽬的⾔語のコアができたことになります。3.3.4節の
デジタル論理シミュレータや 3.3.5節の制約伝播は、それら⾃⾝がちゃんとし
た⾔語になっていて、それぞれ独⾃の基本要素、組み合わせ⼿段、抽象化の⼿
段を持っています。この視点から⾒ると、⼤規模コンピュータシステムに取り
組む技術と新しいコンピュータ⾔語を構築する技術が合流し、計算機科学その
ものが、適切な記述⾔語を構築するという分野にほかならないものとなります。
これから私たちは、⾔語をほかの⾔語によって構築する技術への旅に出発

します。この章では、Lispを基盤として使い、いろいろな評価器を Lispの⼿続
きとして実装していきます。Lispは、記号式を表現し、操作する能⼒のために、
特にこのタスクに向いています。まずは、Lisp⾃⾝の評価器を構築することに
よって、⾔語がどのように実装されるかを理解するための最初の⼀歩を踏み出
します。ここでの評価器によって実装される⾔語は、この本で使っている Lisp
の Scheme⽅⾔のサブセットになります。この章で記述する評価器は、Lispの
特定の⽅⾔に対してのものですが、逐次式コンピュータのプログラムを書くた
めに設計された式指向⾔語の評価器が持つ本質的な構造を持っています (実際
のところ、ほとんどの⾔語処理系は、奥深いところで、⼩さな “Lisp”評価器
を持っているものです)。この評価器は説明と議論のために簡単にしたもので、
製品レベルの Lispシステムにとっては重要な⼀部の機能が省略されています。

それでも、この単純な評価器は、この本のプログラムのほとんどを実⾏するの
に⼗分です。2

評価器を Lispプログラムとしてアクセスできるようにすることの重要な利
点のひとつは、別の評価規則を実装できるようになるということがあります。
これは、評価器プログラムに対する修正としてそれらの規則を記述することに
よって可能になります。この⼒を活⽤できる場所のひとつは、計算モデルが時
間という概念を具体化する⽅法について、より制御しやすくするという場⾯で
す。時間という概念は、第 3章の議論で⾮常に重要なものとなっていました。
そのときは、ストリームを使って実世界の時間の表現とコンピュータの時間と
のつながりを断ち切ることによって、状態と代⼊の複雑さをいくらか緩和でき
ました。しかし、そこでのストリームプログラムは、Schemeの適⽤順序評価と
いう制約があったため、⾯倒なことになることもありました。4.2節では、評価
器に⼿を加えて正規順序評価 (normal-order evaluation)に対応させることで、
よりエレガントなアプローチができるように基層⾔語を変えてしまいます。

4.3節では、もっと⼤胆に⾔語を変更します。ここでは、式がひとつの値だけ
ではなく、複数の値を持つようにします。この⾮決定性計算 (nondeterministic
computing) ⾔語では、式のすべての可能な値を⽣成し、その後で何らかの制約
を満たす値を検索するというプロセスを⾃然に表現できます。計算と時間のモ
デルということで⾔うと、これは時間を “可能な複数の未来”に分岐させ、それ
から適切な時系列を検索するようなものです。この⾮決定性評価器では、複数
の値を管理して検索を⾏うということが、基層となる⾔語の仕組みによって、
⾃動的に⾏われることになります。

4.4節では、論理プログラミング (logic-programming)⾔語を実装します。論
理プログラミング⾔語では、知識を⼊出⼒のある計算としてではなく、関係に
よって表します。こうすると、この⾔語は Lispとは (Lispだけでなく、既存の
どんな⾔語とも)⼤幅に違うものになります。そうでありながら、この論理プ
ログラミング⾔語は本質的には Lisp評価器と同じ構造を持っているというこ
とを、そこで⾒ていきます。

2この評価器に含めなかった機能のうちで最も重要なものは、エラー処理とデバッグ⽀
援の仕組みです。評価器についてのより詳細な議論については、Friedman et al. 1992を
参照してください。この本は、Schemeで次々と評価器を書き、改良していくことによっ
て、さまざまなプログラミング⾔語を実装してみせます。

4.1 メタ循環評価器
ここで作る Lisp評価器は、Lispプログラムとして実装していきます。Lispプ
ログラムを評価するのに、それ⾃⾝ Lispで実装されている評価器を使うとい
うのは、ちょっと考えると循環論のように思えるかもしれません。しかし、評
価というのはプロセスです。そのため、評価プロセスを Lispによって記述す
るというのは妥当なことです。Lispというのは、結局のところ、プロセスを記
述するための道具だからです。3評価する⾔語と同じ⾔語で書かれた評価器は、
メタ循環 (metacircular)であると⾔われます。
メタ循環評価器は、本質的には3.2節で記述した評価の環境モデルを Scheme

によって形式化したものです。モデルには、以下の⼆つの基本部品があったこ
とを思い出しましょう。

1. 組み合わせ (特殊形式以外の複合式)を評価するには、部分式をすべて評
価し、それから演算⼦となる部分式の値を被演算⼦となる部分式の値に
適⽤する。

2. 複合式を引数の集合に適⽤するには、⼿続きの本体を新しい環境で評価
する。この環境を構築するには、⼿続きオブジェクトの環境部分にフレ
ームをひとつ加えて拡張し、その中で⼿続きの仮引数を⼿続き適⽤対象
の引数に束縛する。

これら⼆つの規則は、評価プロセスの本質を記述したものです。環境の中で評
価する式は引数に適⽤する⼿続きに帰着し、それが今度は新しい環境の中で評
価する新しい式に帰着し . . .を繰り返し、環境の中で値が⾒つかる記号や直接
適⽤する基本⼿続きにたどり着くまで続けるというのが、評価プロセスの基本
サイクルです (図 4.1参照)。4

3これは確かにそうなのですが、この評価器では明らかにできない評価プロセスの重
要な側⾯がいくつか残ります。その中でも最も重要なのが、⼿続きがほかの⼿続きを呼
び出し、呼び出し元に返り値を返すメカニズムの詳細です。これらの問題については、第
5章で⾒ていくことにします。そこでは、評価器を単純なレジスタマシンとして実装する
ことによって、評価プロセスをより詳しく調べることになります。

4基本⼿続きを適⽤してもいいということになると、評価器の中で実装すべきものは
何が残るのでしょうか。この評価器の仕事は、⾔語の基本要素を実装することではなく、
基本要素の集合を結合することによって⾔語を形成するための結合組織—組み合わせ⼿
段と抽象化⼿段—を提供するというところにあります。具体的には、

• この評価器は⼊れ⼦の式を扱うことができます。例えば、式 (+ 1 6)の評価は単純
に基本⼿続きを適⽤するだけでできますが、(+ 1 (* 2 3))を扱うにはそれだけでは不

評価器の実装は、評価する式の構⽂ (syntax)を定義する⼿続きに依存しま
す。ここでは、データ抽象化を使って、評価器本体を⾔語の表現に依存しない
ようにしておきます。例えば、代⼊を表すのに set!記号から始まるリストを
使うというように決め打ちするのではなく、抽象述語 assignment?を使って
代⼊をテストし、代⼊の部品にアクセスするには抽象セレクタ assignment-
variableと assignment-valueを使うことにします。式の実装は 4.1.2節で詳
しく説明します。また、⼿続きと環境の表現を規定する演算もあり、これら
については4.1.3節で記述します。例えば、make-procedureは複合⼿続きを構
築し、lookup-variable-valueは変数の値にアクセスし、apply-primitive-
procedureは基本⼿続きを与えられた引数のリストに対して適⽤します。

4.1.1 評価器のコア
評価プロセスは、evalと applyという⼆つの⼿続きの相互作⽤として記述で
きます。

eval

evalは、引数として式と環境を取り、式を分類して評価を振り分けます。eval
の構造は、評価対象の式が持つ構⽂上の型によって場合分けをするという形に
なります。⼿続きに⼀般性を持たせるため、式の型決定を抽象的に表現し、式
の持つさまざまな型の特定の表現に決め打ちしないようにします。式の型はそ
れぞれ、その型かどうかをテストする述語と、構成部品を取り出す抽象アクセ
サを持っています。この抽象構⽂ (abstract syntax)を使うと、同じ評価器を使
⼗分です。基本⼿続き +は引数として数値だけを受け付けるので、式 (* 2 3)を渡され
ても困ってしまうでしょう。評価器の重要な役割のひとつは、組み合わさった⼿続きを
うまく解きほぐして、+の引数として渡される前に (* 2 3)を 6に帰着させるというこ
とです。

• この評価器では変数が使えます。例えば、加算の基本⼿続きは (+ x 1)のような式
を扱うすべがありません。評価器に変数を管理させて、基本⼿続きを呼び出す前に変数
の値を得るようにする必要があります。

• この評価器では複合⼿続きが定義できます。これは、⼿続き定義を管理し、それらの
定義を式評価の中で使うことができるようにし、⼿続きが引数を受け⼊れられるように
するメカニズムを提供するということです。

• この評価器は特殊形式を提供します。特殊形式は、⼿続き呼び出しとは別の評価の
仕⽅をする必要があります。

Eval Apply
Procedure,
Arguments

Expression,
Environment

図 4.1: eval-applyサイクルが明らかにするコンピュータ⾔
語の本質

いつつ別の構⽂⼿続きの集合を使うことによって⾔語の構⽂を変更できるとい
うことがわかりやすくなります。

基本式
• 数値のような⾃⼰評価式については、evalは式そのものを返します。
• evalは、環境内で変数を検索して、変数の値を調べる必要があります。

特殊形式
• クォート式については、evalはクォートの中にある式を返します。
• 変数への代⼊ (あるいは変数の定義)は、再帰的に evalを呼び出し、変
数に関連づける新しい値を計算する必要があります。さらに、環境を修
正して、変数束縛を変更 (または作成)する必要があります。

• if式の部品は特殊な処理をする必要があります。述語が真であれば結果
式を評価し、偽であれば代替式を評価するようにするためです。

• ラムダ式は、ラムダ式の規定する引数と本体、それと評価環境をまとめ
てパッケージ化し、適⽤可能な⼿続きに変形する必要があります。

• begin式は、⼀連の式を出現順に評価する必要があります。
• 場合分け (cond)は、⼊れ⼦の if式に変形し、それから評価します。

組み合わせ
• ⼿続きを適⽤するためには、evalは組み合わせ中の演算⼦部分と被演算
⼦部分を再帰的に評価する必要があります。その結果得られる⼿続きと
引数は applyに渡し、applyが実際の⼿続き適⽤を扱います。

以下に evalの定義を⽰します。
(define (eval exp env)

(cond ((self-evaluating? exp) exp)
((variable? exp) (lookup-variable-value exp env))
((quoted? exp) (text-of-quotation exp))
((assignment? exp) (eval-assignment exp env))
((definition? exp) (eval-definition exp env))
((if? exp) (eval-if exp env))
((lambda? exp) (make-procedure (lambda-parameters exp)

(lambda-body exp)
env))

((begin? exp)
(eval-sequence (begin-actions exp) env))

((cond? exp) (eval (cond->if exp) env))
((application? exp)
(apply (eval (operator exp) env)

(list-of-values (operands exp) env)))
(else
(error "Unknown expression type: EVAL" exp))))

簡単にするために、evalは condを使った場合分けとして実装しています。こ
れの⽋点は、この⼿続きが扱えるのは限られた識別可能な式の型のみで、新し
いものを定義しようと思うと evalの定義を編集しないといけないということ
です。ほとんどの Lisp実装では、式の型によるディスパッチはデータ主導ス
タイルで⾏われています。そのことによって、evalが識別できる新たな式の型
を、evalそのものの定義を修正することなしにユーザーが追加できるように
なっています (練習問題 4.3参照)。

apply

apply は⼆つの引数を取ります。⼿続きと、その⼿続きを適⽤する引数のリ
ストです。applyは⼿続きを⼆種類に分類します。基本演算の場合は、apply-

primitive-procedureを呼んで適⽤します。複合⼿続きの場合は、⼿続きの本
体を構成する式を順番に評価していくことによって適⽤を⾏います。複合⼿続
きの本体を評価する環境は、⼿続きの持つ基本環境を拡張して、⼿続きの仮引
数を⼿続き適⽤対象の引数に束縛するフレームを追加することによって構築し
ます。applyの定義は以下のようになります。
(define (apply procedure arguments)

(cond ((primitive-procedure? procedure)
(apply-primitive-procedure procedure arguments))

((compound-procedure? procedure)
(eval-sequence

(procedure-body procedure)
(extend-environment

(procedure-parameters procedure)
arguments
(procedure-environment procedure))))

(else
(error
"Unknown procedure type: APPLY" procedure))))

⼿続きの引数
evalが⼿続き適⽤を処理する際には、list-of-valuesを使って、⼿続き適⽤
対象の引数のリストを作ります。list-of-valuesは、組み合わせの被演算⼦
を引数に取り、各被演算⼦を評価し、対応する値のリストを返します。5

(define (list-of-values exps env)
(if (no-operands? exps)

'()
(cons (eval (first-operand exps) env)

(list-of-values (rest-operands exps) env))))

5evalの application?節は、明⽰的に list-of-values⼿続きを書くよりも、mapを
使えば (また、operandsがリストを返すよう規定すれば) もっと簡単に書けたところで
す。ここで mapを使わないことにしたのは、評価器がサポートする⾔語が⾼階⼿続きを
持っているとしても、評価器は⾼階⼿続きを使わずに (つまり、⾼階⼿続きを持たない⾔
語でも)実装できるということを強調したかったからです。

条件⽂
eval-ifは、与えられた環境の中で if式の述語部分を評価します。もし結果が
真であれば、eval-ifは結果式 (consequent)を評価し、そうでなければ代替式
(alternative)を評価します。
(define (eval-if exp env)

(if (true? (eval (if-predicate exp) env))
(eval (if-consequent exp) env)
(eval (if-alternative exp) env)))

eval-if内で true?を使うことは、被実装⾔語と実装⾔語の接続という問題を
浮き彫りにします。if-predicateは被実装⾔語によって評価されるため、こ
れはその⾔語での値を返すことになります。インタプリタの述語 true?は、そ
の値を実装⾔語の ifでテストできる値に翻訳します。真であることのメタ循
環表現は、基層となる Schemeの表現と異なるかもしれないからです。6

列
eval-sequenceは、applyが⼿続きの本体中の式の列を評価するときや、eval
が begin式中の式の列を評価するときに使います。引数として式の列と環境を
取り、出現順に式を評価していきます。返り値は最後の式の値です。
(define (eval-sequence exps env)

(cond ((last-exp? exps)
(eval (first-exp exps) env))

(else
(eval (first-exp exps) env)
(eval-sequence (rest-exps exps) env))))

代⼊と定義
以下の⼿続きは変数への代⼊を扱います。evalを呼ぶことによって代⼊する
値を求め、変数とその値を set-variable-value!に渡すことによって指定さ
れた環境に設定します。

6今回は、実装⾔語と被実装⾔語は同じです。ここでの true?の意味をよく考えると、
薬物に頼ることなく、意識を拡張することが可能です。

(define (eval-assignment exp env)
(set-variable-value! (assignment-variable exp)

(eval (assignment-value exp) env)
env)

'ok)

変数の定義も似たようなやり⽅で扱います。7

(define (eval-definition exp env)
(define-variable! (definition-variable exp)

(eval (definition-value exp) env)
env)

'ok)

ここでは、代⼊や定義の値として記号 okを返すことにしています。8

練習問題 4.1: メタ循環評価器が被演算⼦を評価する順番は、左か
ら右なのか、右から左なのか、今のコードを⾒ただけではわから
ないということに注意せよ。評価順は、基層となる Lispのものを
引き継ぐ。つまり、list-of-valuesにある consの引数が左から
右という順で評価されるのであれば、list-of-valuesは被演算⼦
を左から右に評価することになる。consの引数が右から左という
順で評価されるのであれば、list-of-valuesは被演算⼦を右から
左に評価することになる。
基層となる Lispの評価順によらず、被演算⼦を左から右に評価す
るようにしたバージョンの list-of-valuesを書け。また、被演算
⼦を右から左に評価するバージョンの list-of-valuesも書け。

4.1.2 式の表現
この評価器は、2.3.2節で検討した記号微分プログラムと共通するところがあ
ります。どちらのプログラムも記号式の演算を⾏います。どちらのプログラム
でも、複合式の演算結果は、式の各部品に対して演算を⾏い、その結果を式の
型に従って組み合わせることによって得られます。どちらのプログラムでも、

7この defineの実装は、多くの場合正しく動くのですが、内部定義の扱いに関する難
しい問題を無視しています。問題の性質や解決⽅法については4.1.6節で⾒ていきます。

8defineと set!を紹介したときに述べたように、これらの値は Schemeの実装依存
です—つまり、返す値は実装者が決めることができます。

データ抽象化を使って、演算の⼀般的な規則と式の表現⽅法の詳細を切り離し
ています。微分プログラムでは、このことによって、同じ微分⼿続きを使って
前置・中置・後置その他の形式の数式を扱えるようになっていました。評価器
の場合は、評価対象の⾔語の⽂法が、純粋に式の部品を分類し抽出する⼿続き
のみによって決まるという結果になります。
以下は、この⾔語の構⽂の仕様です。
• ⾃⼰評価式は数値と⽂字列しかありません。

(define (self-evaluating? exp)
(cond ((number? exp) true)

((string? exp) true)
(else false)))

• 変数は記号によって表現されます。
(define (variable? exp) (symbol? exp))

• クォート式は (quote <text-of-quotation>)という形式です。9

(define (quoted? exp) (tagged-list? exp 'quote))
(define (text-of-quotation exp) (cadr exp))

quoted?は、リストが指定された記号から始まるかどうかを確認する⼿
続き tagged-list? を使って定義します。
(define (tagged-list? exp tag)

(if (pair? exp)
(eq? (car exp) tag)
false))

• 代⼊は (set! <var> <value>)という形式です。
(define (assignment? exp) (tagged-list? exp 'set!))
(define (assignment-variable exp) (cadr exp))
(define (assignment-value exp) (caddr exp))

92.3.1節で触れたように、評価器はクォート式を quoteから始まるリストとして扱い
ます。これは、クォート式に引⽤符が使われている場合でも同じです。例えば、式'aは、
この評価器によって (quote a)として扱われます。練習問題 2.55参照。

• 定義は以下の形式か、
(define ⟨var⟩ ⟨value⟩)

以下の形式です。
(define (⟨var⟩ ⟨parameter1⟩ . . . ⟨parametern⟩)
⟨body⟩)

後者の形式 (標準⼿続き定義)は、以下のシンタックスシュガーです。
(define ⟨var⟩

(lambda (⟨parameter1⟩ . . . ⟨parametern⟩)
⟨body⟩))

これに対応する構⽂⼿続きは以下の通りです。
(define (definition? exp) (tagged-list? exp 'define))
(define (definition-variable exp)

(if (symbol? (cadr exp))
(cadr exp)
(caadr exp)))

(define (definition-value exp)
(if (symbol? (cadr exp))

(caddr exp)
(make-lambda (cdadr exp) ; 仮引数

(cddr exp)))) ; 本体

• lambda式は、記号 lambdaから始まるリストです。
(define (lambda? exp) (tagged-list? exp 'lambda))
(define (lambda-parameters exp) (cadr exp))
(define (lambda-body exp) (cddr exp))

また、lambda式のコンストラクタも提供します。これは、上記の definition-
valueで使われています。
(define (make-lambda parameters body)

(cons 'lambda (cons parameters body)))

• 条件式は ifで始まり、述語・結果式・ (オプションで)代替式を持ちま
す。式が代替式部分を持たない場合は、代替式として falseを使います。
10

(define (if? exp) (tagged-list? exp 'if))
(define (if-predicate exp) (cadr exp))
(define (if-consequent exp) (caddr exp))
(define (if-alternative exp)

(if (not (null? (cdddr exp)))
(cadddr exp)
'false))

また、if式に対するコンストラクタも提供します。これは、cond->ifが
cond式を if式に変形するのに使われます。
(define (make-if predicate consequent alternative)

(list 'if predicate consequent alternative))

• begin は、式の列を単独の式にパッケージ化します。構⽂演算として、
begin式から実際の列を取り出す演算と、列の最初の式と残りの式を返
すセレクタを⽤意します。11

(define (begin? exp) (tagged-list? exp 'begin))
(define (begin-actions exp) (cdr exp))
(define (last-exp? seq) (null? (cdr seq)))
(define (first-exp seq) (car seq))
(define (rest-exps seq) (cdr seq))

また、cond->ifで使うために、コンストラクタ sequence->expも⽤意し
ます。これは列を単⼀の式に変形するもので、必要であれば beginを使
います。

10述語が falseで代替式が存在しない場合の if式の値は、Schemeでは規定されていま
せん。ここでは falseにすることにしました。変数 trueと falseは、グローバル環境で
束縛することによって、式の中で使えるようにします。4.1.4節参照。

11式のリストに対するこれらのセレクタ—また、これに対応する被演算⼦のリストに
対するセレクタ—は、データ抽象化のためのものではありません。これらは、5.4節の明
⽰的制御評価器の理解を助けるために導⼊した、基本的なリスト演算を覚えやすくする
ための名前です。

(define (sequence->exp seq)
(cond ((null? seq) seq)

((last-exp? seq) (first-exp seq))
(else (make-begin seq))))

(define (make-begin seq) (cons 'begin seq))

• ⼿続き適⽤は、ここまで挙げた式の型に含まれない任意の複合式です。
式の carは演算⼦で、cdrは被演算⼦のリストです。
(define (application? exp) (pair? exp))
(define (operator exp) (car exp))
(define (operands exp) (cdr exp))
(define (no-operands? ops) (null? ops))
(define (first-operand ops) (car ops))
(define (rest-operands ops) (cdr ops))

派⽣式
この⾔語の特殊形式の中には、直接実装するのではなく、ほかの特殊形式を使
った式によって定義できるものもあります。そのひとつの例が condで、⼊れ
⼦の if式として実装できます。例えば、次の式を評価するという問題は、
(cond ((> x 0) x)

((= x 0) (display 'zero) 0)
(else (- x)))

if と begin 式を持つ次の式を評価するという問題に帰着させることができ
ます。
(if (> x 0)

x
(if (= x 0)

(begin (display 'zero) 0)
(- x)))

condをこのように実装することで、評価器を単純にできます。明⽰的に評価
プロセスを規定しなければならない特殊形式の数を減らすことができるから
です。

cond式の部品を抽出する構⽂⼿続きと、cond式を if式に変形する⼿続き
cond->ifを⽤意します。場合分けは、condから始まり、述語-アクション節の
リストを持ちます。節は、述語が記号 elseであれば else節となります。12

(define (cond? exp) (tagged-list? exp 'cond))
(define (cond-clauses exp) (cdr exp))
(define (cond-else-clause? clause)

(eq? (cond-predicate clause) 'else))
(define (cond-predicate clause) (car clause))
(define (cond-actions clause) (cdr clause))
(define (cond->if exp) (expand-clauses (cond-clauses exp)))
(define (expand-clauses clauses)

(if (null? clauses)
'false ; else 節はない
(let ((first (car clauses))

(rest (cdr clauses)))
(if (cond-else-clause? first)

(if (null? rest)
(sequence->exp (cond-actions first))
(error "ELSE clause isn't last: COND->IF"

clauses))
(make-if (cond-predicate first)

(sequence->exp (cond-actions first))
(expand-clauses rest))))))

condのような、構⽂変形として実装するようにした式のことは派⽣式 (derived
expression)と⾔います。let式も同じく派⽣式です (練習問題 4.6参照)。13

練習問題 4.2: Louis Reasonerは、evalの condを並べ替えて、⼿
続き適⽤の節が代⼊の節より前に来るようにしようとしている。

12すべての述語が falseで else節が存在しない場合、Schemeでは cond式の値は規定
されていません。ここでは falseとしています。

13実⽤的な Lispシステムでは、ユーザーが新しい派⽣式を追加し、評価器を修正する
ことなく構⽂変形としての実装を規定できる仕組みを提供しています。そのようなユー
ザー定義の変形は マクロ (macro)と呼ばれます。単純なマクロ定義の仕組みを追加する
ことは簡単なのですが、そうすると⾔語が名前の衝突という厄介な問題を持つことにな
ります。そのような問題のないマクロ定義の仕組みについては多くの研究がなされてい
ます。例えば、Kohlbecker 1986, Clinger and Rees 1991, Hanson 1991を参照してくだ
さい。

彼は、こうするとインタプリタの効率がよくなると⾔う。プログ
ラムは通常、代⼊や定義などより適⽤を多く含んでいるのだから、
修正版の evalは、元の evalよりも式の型を識別するまでにチェ
ックする節の数が少なくなると⾔うのだ。

a. Louisの計画はどこがまずいのだろうか (ヒント：Louisの評
価器は、式 (define x 3)をどのように処理することになる
だろうか)。

b. Louisは計画がうまくいかないのでイライラしている。彼は、
何が何でも、ほかの⼤部分の型に対するチェックよりも先に、
評価器に⼿続き適⽤を認識させたいと思っている。評価対象
の⾔語の構⽂を変更し、⼿続き適⽤が callで始まるようす
ることで、彼の⼿伝いをせよ。例えば、(factorial 3)と書
いていたところで、変更後は (call factorial 3)と書かな
ければならなくなる。(+ 1 2)は、(call + 1 2)と書かなけ
ればならない。

練習問題 4.3: ディスパッチをデータ主導スタイルで⾏うように
evalを書き直せ。これを 練習問題 2.73のデータ主導型微分⼿続き
と⽐較せよ (式の型としては、複合式の carを使ってよい。この節
で実装した構⽂ではそれが適切である)。
練習問題 4.4: 第 1章で出てきた特殊形式である andと orの定義
を思い出そう。

• and: 式は左から右へと評価される。式のいずれかが偽と評価
されたら偽を返す。残りの式はすべて評価されない。すべて
の式が真の値として評価されるなら、最後の式の評価値を返
す。式がなければ真を返す。

• or: 式は左から右へと評価される。式のいずれかが真の値と
して評価されたらその値を返す。残りの式はすべて評価され
ない。すべての式が偽と評価されるか、式が存在しない場合
は、偽を返す。

適切な構⽂⼿続きと、評価⼿続き eval-and, eval-orを定義し、評
価器の新たな特殊形式として andと orを組み込め。また、別のや
り⽅として、andと orを派⽣式として実装する⽅法を⽰せ。

練習問題 4.5: Schemeには、cond節の別の構⽂として、(<test>
=> <recipient>)というものがある。これは、⟨test⟩が真の値とし
て評価される場合、⟨recipient⟩が評価される。その値は 1引数の
⼿続きである必要がある。そして、この⼿続きが ⟨test⟩という値
で呼び出され、その結果が cond式の値として返される。例えば、
(cond ((assoc 'b '((a 1) (b 2))) => cadr)

(else false))

は 2を返す。この拡張構⽂をサポートするよう condの処理を修正
せよ。
練習問題 4.6: let式は派⽣式である。以下は、
(let ((⟨var1⟩ ⟨exp1⟩) . . . (⟨varn⟩ ⟨expn⟩))
⟨body⟩)

次のものと等価だからだ。
((lambda (⟨var1⟩ . . . ⟨varn⟩)
⟨body⟩)

⟨exp1⟩
. . .

⟨expn⟩)

let式の評価を、上記のような組み合わせの評価に帰着させる構⽂
変形 let->combinationを実装せよ。また、evalに適切な節を追
加し、let式を扱えるようにせよ。
練習問題 4.7:
let*は letに似ているが、let*変数の束縛は左から右へ順番に⾏
われ、それぞれの束縛は、それまでの束縛がすべて⾒える環境の
中で⾏われるというところが違う。例えば、
(let* ((x 3) (y (+ x 2)) (z (+ x y 5)))

(* x z))

は 39を返す。let*式を⼊れ⼦の let式の集合として書き直す⽅
法を説明せよ。また、この変形を実⾏する⼿続き let*->nested-
letsを書け。もしすでに letを実装していて (練習問題 4.6)、let*
を扱えるように評価器を拡張したいとしたら、以下のアクション
を持つ節を evalに追加することで⾜りるだろうか。

(eval (let*->nested-lets exp) env)

それとも、⾮派⽣式によって明⽰的に let*を拡張する必要がある
だろうか。
練習問題 4.8: “名前つき let”は letの変種で、以下の形式を持つ。
(let ⟨var⟩ ⟨bindings⟩ ⟨body⟩)

⟨bindings⟩と ⟨body ⟩は通常の letと同じである。違うのは、⟨body ⟩
内部で ⟨var⟩が、本体が ⟨body ⟩で引数が ⟨bindings⟩内の変数であ
る⼿続きに束縛されるというところだ。このため、⟨var⟩という名
前の⼿続きを呼び出すことによって、繰り返し ⟨body ⟩を実⾏でき
る。例えば、反復フィボナッチ⼿続き (1.2.2節)は、名前つき let
を使って次のように書き直すことができる。
(define (fib n)

(let fib-iter ((a 1)
(b 0)
(count n))

(if (= count 0)
b
(fib-iter (+ a b) a (- count 1)))))

練習問題 4.6の let->combinationを修正し、名前つき letもサポ
ートするようにせよ。
練習問題 4.9: 多くの⾔語では、do, for, while, untilのようない
ろいろな反復構造をサポートしている。Schemeでは、反復処理は
通常の⼿続き呼び出しによって表現できるので、特別な反復構造
を⼊れても計算能⼒が本質的に上がるわけではない。⼀⽅で、そ
のような構造を持っていると便利なことも多い。いくつかの反復
構造を設計し、使⽤例を⽰し、それらを派⽣式として実装するや
り⽅を⽰せ。
練習問題 4.10: データ抽象化を使うことで、評価対象⾔語の個別
の⽂法とは独⽴した eval⼿続きを書くことができた。そのことを
⽰すために、evalと applyを変更せずに、この節の⼿続きを変更
することによって Schemeの新しい⽂法を設計・実装せよ。

4.1.3 評価器のデータ構造
評価器を実装するときには、式の外部⽂法の定義だけではなく、評価器がプロ
グラム実⾏の⼀部として内部で操作するデータ構造も定義する必要がありま
す。これには、⼿続きや環境の表現、真と偽の表現のようなものがあります。

述語のテスト
条件式については、明⽰的な falseオブジェクト以外はすべて trueとして受
け⼊れます。
(define (true? x) (not (eq? x false)))
(define (false? x) (eq? x false))

⼿続きの表現
基本⼿続きを扱うときには、以下の⼿続きが利⽤できるものとします。

• (apply-primitive-procedure <proc> <args>)
これは、与えられた基本⼿続きをリスト ⟨args⟩中の引数の値に適⽤し、
その結果を返します。

• (primitive-procedure? <proc>)
これは、⟨proc⟩が基本⼿続きかどうかテストします。

これらの基本⼿続きを扱う仕組みについては、4.1.4節でさらに詳しく説明し
ます。
複合⼿続きは、コンストラクタ make-procedureを使って、仮引数・⼿続

き本体・環境によって構築します。
(define (make-procedure parameters body env)

(list 'procedure parameters body env))
(define (compound-procedure? p)

(tagged-list? p 'procedure))
(define (procedure-parameters p) (cadr p))
(define (procedure-body p) (caddr p))
(define (procedure-environment p) (cadddr p))

環境に対する演算
評価器には環境を操作する演算が必要となります。3.2節で説明したように、環
境はフレームの列で、各フレームは変数を対応する値に関連づける束縛テーブ
ルです。環境を操作するには以下の演算を使います。

• (lookup-variable-value <var> <env>)
これは、環境 ⟨env ⟩内で記号 ⟨var⟩に束縛されている値を返します。変
数が束縛されていない場合、エラーを発⽣させます。

• (extend-environment <variables> <values> <base-env>)
これは、リスト ⟨variables⟩の記号がリスト ⟨values⟩の対応する要素に束
縛されている新しいフレームひとつからなる新しい環境を返します。環
境 ⟨base-env ⟩が外側の環境となります。

• (define-variable! <var> <value> <env>)
これは、環境 ⟨env ⟩の⼀つ⽬のフレームに、変数 ⟨var ⟩と値 ⟨value⟩を関
連づける新たな束縛を追加します。

• (set-variable-value! <var> <value> <env>)
これは、環境 ⟨env ⟩中の変数 ⟨var⟩の束縛を変更して、変数を値 ⟨value
⟩に束縛し直します。変数が未束縛であればエラーを発⽣させます。

これらの演算を実装するために、環境はフレームのリストとして表現すること
にします。リストの cdrが外側の環境となります。空の環境は、単純に空リス
トです。
(define (enclosing-environment env) (cdr env))
(define (first-frame env) (car env))
(define the-empty-environment '())

環境のそれぞれのフレームはリストのペアとして表現します。ひとつはそのフ
レームで束縛される変数のリストで、もうひとつは関連づけられる値のリスト
です。14

(define (make-frame variables values)
(cons variables values))

14ここから先のコードでは、フレームはデータ抽象化されていません。set-variable-
value!と define-variable!は、set-car!を使ってフレームの値を直接変更していま
す。このフレーム⼿続きの⽬的は、環境操作の⼿続きを読みやすくすることです。

(define (frame-variables frame) (car frame))
(define (frame-values frame) (cdr frame))
(define (add-binding-to-frame! var val frame)

(set-car! frame (cons var (car frame)))
(set-cdr! frame (cons val (cdr frame))))

変数を値に関連づける新しいフレームによって環境を拡張するには、変数のリ
ストと値のリストからなるフレームを作り、それを環境に追加します。変数の
数と値の数が⼀致しなければエラーを発⽣させます。
(define (extend-environment vars vals base-env)

(if (= (length vars) (length vals))
(cons (make-frame vars vals) base-env)
(if (< (length vars) (length vals))

(error "Too many arguments supplied" vars vals)
(error "Too few arguments supplied" vars vals))))

環境内の変数を探すには、⼀つ⽬のフレームの変数リストを⾛査します。求め
る変数が⾒つかれば、値リスト内の対応する要素を返します。現在のフレーム
で変数が⾒つからなければ、外側の環境、さらにその外側の環境と検索してい
きます。空の環境に達したら、“未束縛変数”エラーを発⽣させます。
(define (lookup-variable-value var env)

(define (env-loop env)
(define (scan vars vals)

(cond ((null? vars)
(env-loop (enclosing-environment env)))

((eq? var (car vars)) (car vals))
(else (scan (cdr vars) (cdr vals)))))

(if (eq? env the-empty-environment)
(error "Unbound variable" var)
(let ((frame (first-frame env)))

(scan (frame-variables frame)
(frame-values frame)))))

(env-loop env))

指定した環境の中で変数を新しい値に設定するには、lookup-variable-value
と同じように変数を⾛査し、⾒つかったら対応する値を変更します。

(define (set-variable-value! var val env)
(define (env-loop env)

(define (scan vars vals)
(cond ((null? vars)

(env-loop (enclosing-environment env)))
((eq? var (car vars)) (set-car! vals val))
(else (scan (cdr vars) (cdr vals)))))

(if (eq? env the-empty-environment)
(error "Unbound variable: SET!" var)
(let ((frame (first-frame env)))

(scan (frame-variables frame)
(frame-values frame)))))

(env-loop env))

変数を定義するには、その変数の束縛を⼀つ⽬のフレームで探し、束縛がす
でにあればその束縛を変更します (これは set-variable-value!と同じです)。
束縛がなければ、⼀つ⽬のフレームに追加します。
(define (define-variable! var val env)

(let ((frame (first-frame env)))
(define (scan vars vals)

(cond ((null? vars)
(add-binding-to-frame! var val frame))

((eq? var (car vars)) (set-car! vals val))
(else (scan (cdr vars) (cdr vals)))))

(scan (frame-variables frame) (frame-values frame))))

ここで記述した⼿法は、環境を表現する数多くの⽅法のひとつにすぎません。
データ抽象化を使って、どういう表現を選択するかという細かいところを評価
器のほかの部分から分離させたので、やろうと思えば環境の表現⽅法を変える
こともできます (練習問題 4.11参照)。製品品質の Lispシステムでは、評価器
の環境演算速度 (特に、変数探索の速度)がシステムの性能に⼤きく影響しま
す。ここで記述した表現は、概念的には単純なのですが、効率的ではないので
普通は製品システムで使われることはないでしょう。15

15この表現 (練習問題 4.11のバージョンも同様です)の⽋点は、ある変数の束縛を⾒つ
けるのに、評価器がたくさんのフレームを⾒ていかないといけないことがあるというと
ころです (このようなアプローチは深い束縛 (deep binding)と呼ばれます)。この⾮効率
さを防ぐ⽅法のひとつとしてレキシカルアドレッシング (lexical addressing)という戦略

練習問題 4.11: フレームをリストのペアとして表現するのではな
く、束縛のリストとして表現することもできる。この場合、それぞ
れの束縛は名前と値のペアとなる。環境演算を書き直し、この表
現を使うようにせよ。
練習問題 4.12: ⼿続き set-variable-value!, define-variable!,
lookup-variable-valueは、環境構造を順にたどる、より抽象的な
⼿続きによって表現できる。共通パターンを取り出した抽象化を
定義し、それらの抽象化によって上記三つの⼿続きを再定義せよ。
練習問題 4.13: Schemeでは、defineによって新しい束縛を作るこ
とができる⼀⽅で、束縛を削除する⽅法は提供していない。与えら
れた記号の束縛を評価環境から削除する特殊形式 make-unbound!
を実装せよ。この問題の仕様は完全ではない。例えば、削除を⾏う
のは環境の⼀つ⽬のフレームだけでよいのだろうか。仕様を完成
させ、⾏った選択について根拠を述べよ。

4.1.4 評価器をプログラムとして実⾏する
評価器があれば、Lisp式が評価されるプロセスを (Lispによって)記述したも
のを⼿にしたことになります。評価器をプログラムとして表現することの利点
のひとつは、プログラムを実⾏できるということです。こうすると、Lispの中
で実⾏することによって、Lisp⾃⾝が式を評価する⽅法の実⾏モデルが得られ
ます。これは、評価規則を実験する枠組みの役割を果たします。この章の後の
ほうでは、そのような実験を⾏います。
評価器プログラムが式を簡約していくと、最終的には基本⼿続きの適⽤に

なります。そのため、評価器を実⾏するのに必要なのは、基本⼿続きの適⽤を
モデル化するために基層となる Lispシステムを呼び出す仕組みを作ることだ
けです。
基本⼿続きの適⽤を評価したときに、evalが applyに渡すオブジェクトを

⾒つけられるようにそれぞれの基本⼿続きの名前に対して束縛が存在する必要
があります。そのため、グローバル環境を作成し、評価する式に出てくる可能
性のある基本⼿続きの名前にそれぞれ特定のオブジェクトを関連づけます。グ
ローバル環境は、評価する式の中で使えるように、記号 true, falseの束縛も
含んでいます。
を使うというものがあります。これについては5.5.6節で検討します。

(define (setup-environment)
(let ((initial-env

(extend-environment (primitive-procedure-names)
(primitive-procedure-objects)
the-empty-environment)))

(define-variable! 'true true initial-env)
(define-variable! 'false false initial-env)
initial-env))

(define the-global-environment (setup-environment))

基本⼿続きを表現する⽅法は、どのようなものでもかまいません。apply が
primitive-procedure?を使って識別でき、apply-primitive-procedureを使
って適⽤できればそれで⼗分です。ここでは、基本⼿続きはリストとして表現
することにします。リストは記号 primitiveから始まり、その基本⼿続きを実
装している基層 Lispの⼿続きを含むものとします。
(define (primitive-procedure? proc)

(tagged-list? proc 'primitive))
(define (primitive-implementation proc) (cadr proc))

setup-environment は、基本⼿続きの名前と実装⼿続きをリストから取りま
す。16

(define primitive-procedures
(list (list 'car car)

(list 'cdr cdr)
(list 'cons cons)
(list 'null? null?)
⟨more primitives⟩ ))

(define (primitive-procedure-names)
(map car primitive-procedures))

(define (primitive-procedure-objects)

16基層の Lispで定義されている⼿続きは、すべてメタ循環評価器の基本⼿続きとして
使うことができます。この評価器に組み込む基本⼿続きの名前は、基層の Lispでの実装
の名前と同じである必要はありません。ここで名前が同じになっているのは、メタ循環
評価器が実装しているのが Schemeそのものだという理由です。つまり、例えば (list
'first car)や (list 'square (lambda (x) (* x x)))を primitive-proceduresの
リストに含めるということも可能です。

(map (lambda (proc) (list 'primitive (cadr proc)))
primitive-procedures))

基本⼿続きを適⽤するには、基層 Lispシステムを使って実装⼿続きを引数に
適⽤するだけです。17

(define (apply-primitive-procedure proc args)
(apply-in-underlying-scheme
(primitive-implementation proc) args))

メタ循環評価器を⼿軽に動かせるようにするため、基層の Lispシステムの対
話型評価環境 (REPL)をモデル化したドライバループ (driver loop)を⽤意しま
す。これは、プロンプト (prompt)を表⽰し、⼊⼒式を読み込み、その式をグロ
ーバル環境の中で評価し、その結果を表⽰するというものです。表⽰する結果
の前には出⼒プロンプト (output prompt)を置いて、式の値をほかの出⼒から
区別できるようにします。18

(define input-prompt ";;; M-Eval input:")
(define output-prompt ";;; M-Eval value:")
(define (driver-loop)

(prompt-for-input input-prompt)
(let ((input (read)))

(let ((output (eval input the-global-environment)))
(announce-output output-prompt)
(user-print output)))

17apply-in-underlying-schemeは、以前の章で使ってきた apply⼿続きです。メタ
循環評価器の apply⼿続き (4.1.1節)は、この基本⼿続きの動作をモデル化したもので
す。applyという名前のものが⼆つあると、メタ循環評価器を実⾏する際に技術的な問題
が起こります。これは、メタ循環評価器の applyを定義すると、基本⼿続きの定義が隠
されてしまうことによります。回避する⽅法のひとつとしては、メタ循環の applyを別
の名前にして、基本⼿続きの名前との衝突しないようにするというものがあります。こ
こではそうせずに、メタ循環の applyを定義する前に以下のようにして基層の applyへ
の参照を保存してあると仮定しています。
(define apply-in-underlying-scheme apply)

こうすると、オリジナルの applyに別の名前でアクセスできることになります。
18基本⼿続き readはユーザーからの⼊⼒を待ち、⼊⼒された次の完全な式を返します。
例えば、ユーザーが (+ 23 x)と⼊⼒した場合、readは記号 +、数値 23、記号 xを含む
3要素のリストを返します。ユーザーが'xと⼊⼒した場合、readは記号 quoteと記号 x
を含む 2要素のリストを返します。

(driver-loop))
(define (prompt-for-input string)

(newline) (newline) (display string) (newline))
(define (announce-output string)

(newline) (display string) (newline))

ここでは、user-printという特別な表⽰⼿続きを使い、複合⼿続きの環境部分
を表⽰しないようにしています。環境部分はとても⻑いリストになったり、循
環を含んでいたりするかもしれないからです。
(define (user-print object)

(if (compound-procedure? object)
(display (list 'compound-procedure

(procedure-parameters object)
(procedure-body object)
'<procedure-env>))

(display object)))

これで、後はグローバル環境を初期化してドライバループを開始するだけで、
評価器を実⾏できます。以下は対話の例です。
(define the-global-environment (setup-environment))
(driver-loop)

;;; M-Eval input:
(define (append x y)

(if (null? x)
y
(cons (car x) (append (cdr x) y))))

;;; M-Eval value:
ok
;;; M-Eval input:
(append '(a b c) '(d e f))
;;; M-Eval value:
(a b c d e f)

練習問題 4.14: Eva Lu Atorと Louis Reasonerはそれぞれメタ循
環評価器の実験をしていた。Evaは mapの定義を⼊⼒し、それを

使うテストプログラムを実⾏した。テストは問題なくうまくいっ
た。⼀⽅、Louisはシステム版の mapをメタ循環評価器の基本⼿続
きとして組み込んだ。しかし、それを動かそうとすると、ひどいこ
とになった。Evaの mapはうまくいくのに Louisの mapは失敗す
る理由を説明せよ。

4.1.5 プログラムとしてのデータ
⽐喩を使うと、Lisp式を評価する Lispプログラムについて考える助けになる
かもしれません。プログラムの意味についての操作的な⾒⽅のひとつとして、
プログラムは (無限に⼤きいかもしれない) 抽象機械の記述であるというもの
があります。例えば、階乗を計算する⾒慣れたプログラムについて考えてみま
しょう。
(define (factorial n)

(if (= n 1) 1 (* (factorial (- n 1)) n)))

このプログラムのことを機械と考えて、その機械には減算・乗算・等価テスト
という部品があり、また切り替えスイッチと別の階乗機械が⼊っているとい
うように考えることもできます (この階乗機械は、別の階乗機械を中に含んで
いるため無限です)。図 4.2はこの階乗機械の流れ図で、部品の配線を⽰してい
ます。
同じように、評価器についても、機械の記述を⼊⼒として取る⾮常に特殊

な機械として考えることができます。この⼊⼒をもとに、評価器は記述された
機械の動作を模倣するよう⾃分⾃⾝を設定します。例えば、この評価器に図
4.3のような factorialの記述を与えると、評価器は階乗の計算ができるよう
になるということです。
この視点から⾒ると、この評価器は万能機械 (universal machine)であると

いえます。この機械は、Lispプログラムとして記述されているほかの機械につ
いて、それらの動作の真似をすることができます。19 これは特筆すべきことで

19模倣対象の機械が Lispで記述されているということは本質ではありません。この評
価器に、ほかの⾔語 (例えば C⾔語)の評価器としてふるまう Lispプログラムを与える
と、この Lisp評価器は C⾔語評価器の動作を模倣することになり、つまり C⾔語のプ
ログラムとして記述されたどんな機械の動作も模倣できるということになります。同じ
ように、C⾔語で Lisp評価器を書くと、どんな Lispプログラムでも実⾏できる C⾔語
プログラムができます。ここで本質的なことは、どんな評価器でも、ほかの評価器の動
作を模倣できるということです。このため、“何が原理上計算可能なのか” (時間とメモ

図 4.2: 抽象機械として⾒た階乗プログラム

す。例えば、電⼦回路について、同じような評価器を想像できるでしょうか。
その評価器は、例えばフィルタのような、ほかの回路の設計図をコード化した
信号を⼊⼒として取る回路ということになるでしょう。この⼊⼒を与えられる
と、この回路評価器はまるで同じ仕様を持つフィルタのようにふるまうことに
なります。そのような万能電⼦回路というものは、ほとんど想像もできないぐ
リの実⽤性は度外視して)という概念は、⾔語やコンピュータとは独⽴のもので、根底
にある計算可能性 (computability)という概念を反映しています。このことを最初にはっ
きりと⽰したのは Alan M. Turing (1912-1954)です。彼の 1936年の論⽂は理論計算機
科学の基礎を築きました。この論⽂で、チューリングは単純な計算モデル —現在チュー
リングマシン (Turing machine)として知られている—を提⽰し、任意の “実効的処理”
はそのような機械のプログラムとして定式化できると主張しました (この主張は、チャ
ーチ・チューリングのテーゼ (Church-Turing thesis)として知られています)。次に、チ
ューリングは万能機械、つまりチューリングマシン⽤プログラムの評価器としてふるま
うチューリングマシンを実装しました。彼はこの枠組みを使って、チューリングマシン
によって計算できない (練習問題 4.15参照)、つまり “実効的な処理”として定式化でき
ない良設定問題が存在することを実証しました。チューリングはさらに、実⽤的な計算
機科学への基礎的な貢献も続けました。例えば、彼は汎⽤⽬的サブルーチンを使ってプ
ログラミングを構造化するという考え⽅を発明しました。チューリングの伝記について
はHodges 1983を参照してください。

(define (factorial n)
  (if (= n 1)
      1
      (* (factorial (- n 1)) n)))

eval6 720

図 4.3: 階乗マシンの動作を模倣する評価器

らい複雑なものです。プログラムの評価器が⽐較的単純なプログラムであると
いうことは注⽬すべきことです。20

評価器のもうひとつの特筆すべき側⾯は、プログラミング⾔語によって操
作されるデータオブジェクトとプログラミング⾔語そのものとをつなぐ橋の役
割を果たすというところです。ここで、(Lispで実装された) 評価器プログラム
が動いていて、ユーザーがその評価器に式を⼊⼒して、結果を⾒ているところ
を想像してみてください。ユーザーの視点からは、(* x x)のような⼊⼒式は
プログラミング⾔語の式で、評価器が実⾏するべきものです。しかし、評価器
の視点からは、式は明確に定義されたルール集合に従って操作するべき単純な
リスト (この場合、*, x, xという三つの記号からなるリスト)ということになり
ます。
ユーザーにとってのプログラムは評価器にとってはデータということにな

りますが、これは必ずしも混乱するようなことではありません。実際のところ、
この違いを無視して、データオブジェクトを Lispの式として明⽰的に評価す
る能⼒をユーザーに与えるために、eval⼿続きをプログラム中で使えるよう

20単純な⼿続きによって実装された評価器が、評価器そのものよりも複雑なプログラ
ムの動作を模倣できるということが、直感に反すると思う⼈もいます。万能評価機械が
存在するということは、計算というものの深遠ですばらしい性質を表しています。再帰
理論 (Recursion theory)という数理論理学の⼀部⾨では、計算の論理的限界を扱ってい
ます。ダグラス・ホフスタッター (Douglas Hofstadter)の美しい本ゲーデル、エッシャ
ー、バッハ (Gödel, Escher, Bach)では、これらの考えのいくつかについて探求してい
ます (Hofstadter 1979)。

421



にすることが便利であることもあります。Lisp⽅⾔の多くは、引数として式と
環境を取り、その環境について式を評価する基本⼿続きの evalを提供してい
ます。21 このため、
(eval '(* 5 5) user-initial-environment)

と
(eval (cons '* (list 5 5)) user-initial-environment)

は、両⽅とも 25を返します。22

練習問題 4.15: 1引数の⼿続き pとオブジェクト aについて、式
(p a)が (エラーメッセージを出して終了したり、無限に動き続け
るのではなく)値を返すとき、pは aについて “停⽌”するという。
任意の⼿続き pと任意のオブジェクト aに対して、pが aについ
て停⽌するかどうかを正確に決定するような⼿続き halts?という
ものを書くことは不可能であることを⽰せ。以下のように推論せ
よ。もしそのような⼿続き halts?が存在するとしたら、次のよう
なプログラムが実装できるはずだ。
(define (run-forever) (run-forever))
(define (try p)

(if (halts? p p) (run-forever) 'halted))

ここで、式 (try try)を評価した場合について考え、起こりうる
結果 (停⽌するにしても、無限に動き続けるにしても)がすべて、
意図する halts?の動作と⽭盾することを⽰せ。23

21警告：この基本⼿続きの evalは、4.1.1節で実装した eval⼿続きと同⼀のものでは
ありません。というのは、この基本⼿続きの evalは実際の Scheme環境を使うもので、
4.1.3節で構築したようなサンプル環境構造を使うわけではないからです。これらの実際
の環境はユーザーが通常のリストとして操作することはできず、evalやほかの特別な演
算によってアクセスする必要があります。同じように、以前の章で出てきた基本⼿続き
の applyも、メタ循環 applyと同⼀のものではありません。基本⼿続きの applyは、実
際の Scheme⼿続きを使うもので、4.1.3節や 4.1.4節で構築した⼿続きオブジェクトを
使うわけではないからです。

22Schemeの mit実装には evalが含まれていて、また user-initial-environmentと
いう、ユーザーの⼊⼒式が評価される初期環境に束縛された記号も含まれています。

23ここでは halts? が⼿続きオブジェクトを与えられるものと規定していますが、
halts?が⼿続きのテキストとその環境にアクセスできるとしても、依然としてこの推
論が成り⽴つということに注意してください。これはチューリングの有名な停⽌性問題

4.1.6 内部定義
評価の環境モデルとメタ循環評価器は、定義を順に実⾏し、環境のフレームを
⼀定義分ずつ拡張していきます。インタラクティブなプログラム開発では、プ
ログラマは⼿続きの適⽤と新しい⼿続きの定義を⾃由に⾏ったり来たりする必
要があるので、これは特に便利です。しかし、(1.1.8節で導⼊した)ブロック構
造を実装するために使う内部定義についてよく考えると、名前ごとに環境を拡
張するというのは、局所変数を定義する⽅法としてあまり優れているとはいえ
ないのではないかと気づくと思います。
次のような、内部定義のある⼿続きについて考えてみましょう。

(define (f x)
(define (even? n) (if (= n 0) true (odd? (- n 1))))
(define (odd? n) (if (= n 0) false (even? (- n 1))))
⟨rest of body of f⟩)

ここでの意図は、⼿続き even?の本体に含まれる名前 odd?が、even?の後に
定義されている⼿続き odd?を参照するようにするということです。名前 odd?
のスコープは fの本体全体であって、fの本体の中で odd?の定義のある場所
より後ろに限定されるわけではありません。また、odd?⾃⾝も even?を使って
定義されているということを考えると—つまり、even?と odd?は相互再帰⼿
続きということになります—⼆つの defineのちゃんとした解釈⽅法としては、
even?と odd?という名前が同時に環境に追加されているかのように⾒なす以
外にないということがわかります。より⼀般的には、ブロック構造では、局所
的な名前のスコープは、defineが評価されている⼿続きの本体全体だという
ことです。
私たちのインタプリタは、たまたま fの呼び出しを正しく評価できるよう

になっていますが、そうなっているのは “偶然の”理由のためです。内部⼿続き
の定義は最初に来るので、これらの⼿続きへの呼び出しは、これらの⼿続きが
すべて定義される前には評価されないからです。そのため、even?が実⾏され
るときには odd?はすでに定義されていることになります。実際のところ、内
部定義が本体の先頭に来て、定義された変数に対する値の式の評価が定義され
た変数をひとつも実際には使わない⼿続きの場合、ここでの逐次評価メカニズ
ムは同時定義を直接実装したメカニズムと同じ結果を返します (これらの制約
に従わず、逐次定義が同時定義と等価にならないような⼿続きの例については、
(Halting Theorem)で、これが計算不可能 (non-computable)な問題、つまり計算⼿続き
として実⾏不可能な良設定問題の最初の明確な例となりました。

練習問題 4.19を参照してください)。24

しかし、内部定義が真に同時スコープを持つように定義を扱う単純な⽅法
があります。現在の環境にあるすべての局所変数を、値の式を評価する前に作
ってしまうのです。その⽅法のひとつとしては、lambda式に対する構⽂変形を
使うというものがあります。lambda式の本体を評価する前に、本体内の内部
定義を “⾛査”し、削除してしまうのです。内部で定義された変数は letによ
って作られ、それから代⼊によってそれらの持つ値に設定されることになりま
す。例えば、以下の⼿続きは
(lambda ⟨vars⟩

(define u ⟨e1⟩)
(define v ⟨e2⟩)
⟨e3⟩)

以下のように変形されます。
(lambda ⟨vars⟩

(let ((u '*unassigned*)
(v '*unassigned*))

(set! u ⟨e1⟩)
(set! v ⟨e2⟩)
⟨e3⟩))

ここで、*unassigned*は特別な記号で、未割り当ての変数の値を使おうとし
たらエラーを発⽣させるものです。
内部定義を⾛査する以外の戦略のひとつは、練習問題 4.18に⽰されていま

す。上記の変形とは違って、こちらは定義された変数の値はそれらの変数の値
を使わないでも評価できるという制約を強制するものです。25

24第 1章のFootnote 28で “責任を負いかねます”と書いたのは、プログラムにこの評
価メカニズムに依存してほしくないというのがその理由です。内部⼿続きが最初に来て、
定義が評価されている間にお互いを使わないと決めてあるため、Schemeの ieee標準を
実装する際には、これらの定義を評価するメカニズムに選択の余地があります。どの評価
ルールを選び、どの評価ルールを選ばないかということは、“不適格な”プログラムを解
釈する場合以外には影響のない⼩さな問題のようにここでは思えるかもしれません。し
かし、内部定義の同時スコープモデルに移⾏することによって、そうしない場合にコンパ
イラの実装で起こるいくつかの厄介な問題を避けられるようになるということを、5.5.6
節では⾒ていきます。

25Schemeの ieee標準では、この制約に従うのはプログラマの責任で、実装側でそれを
強制する責任はないと規定しているので、実装戦略に幅を持たせています。mit Scheme

練習問題 4.16: この練習問題では、上で説明した内部定義の⼿法
を実装する。評価器は letをサポートしているとする (練習問題
4.6参照)。

a. lookup-variable-value (4.1.3節)を変更し、⾒つけた値が
記号 *unassigned*であればエラーを発⽣させるようにせよ。

b. ⼿続きの本体を引数に取り、上記の変形を施すことによ
って内部⼿続きを持たない等価な⼿続きを返す scan-out-
definesを書け。

c. scan-out-defines をインタプリタの make-procedure また
は procedure-body (4.1.3節参照)に組み込め。どちらに組み
込むのがよいだろうか。また、それはなぜか。

練習問題 4.17: 本⽂の⼿続き中の式 ⟨e3 ⟩を評価しているときに有
効である環境の図を描き、定義が逐次的に解釈される場合と定義
を上記のように掃き出す場合とでそれぞれどのようになるか⽐較
せよ。変形したプログラムでは、なぜフレームがひとつ余分にあ
るのだろうか。環境構造のこの違いが正しいプログラムのふるま
いには決して影響を及ぼさない理由を説明せよ。インタプリタに
余分なフレームを構築することなしに “同時”スコープルールを実
装させる⽅法を設計せよ。
練習問題 4.18: 定義を掃き出す別の戦略として、本⽂中の例を以
下のように変形するものを考える。
(lambda ⟨vars⟩

(let ((u '*unassigned*) (v '*unassigned*))
(let ((a ⟨e1⟩) (b ⟨e2⟩))

(set! u a)
(set! v b))

⟨e3⟩))

ここで、a, bはインタプリタの作成した新しい変数名を表現してい
るものとする。これらはユーザーのプログラムには現れない。3.5.4
節の solve⼿続きについて考える。

を含め、いくつかの Scheme実装は、上に⽰した変形を使っています。そのため、この制
約に従わないプログラムでも、そのような実装では実際には動くこともあります。

(define (solve f y0 dt)
(define y (integral (delay dy) y0 dt))
(define dy (stream-map f y))
y)

この⼿続きは、この練習問題で⽰したように内部定義を掃き出し
た場合に、うまく動くだろうか。本⽂中で⽰したように掃き出し
た場合はどうだろうか。説明せよ。
練習問題 4.19: Ben Bitdiddle, Alyssa P. Hacker, Eva Lu Atorは、
以下の式を評価したときの結果として望ましいものは何かという
ことで議論している。
(let ((a 1))

(define (f x)
(define b (+ a x))
(define a 5)
(+ a b))

(f 10))

Benは、defineに逐次ルールを使って結果を得るべきだと主張す
る。bは 11と定義され、その次に aは 5と定義されるので、結果
は 16であるべきだということだ。Alyssaは、相互再帰には内部⼿
続き定義の同時スコープルールが必要で、⼿続きの名前とほかの
名前で扱いを変えるのは不合理だと反対する。このため、彼⼥は練
習問題 4.16で実装したメカニズムを⽀持している。これを使うと、
bの値を計算する時点で、aは未割り当てということになる。よっ
て、Alyssaの⾒⽅では、⼿続きはエラーを発⽣させるべきだとい
うことだ。Evaの意⾒はどちらとも違う。彼⼥は、aと bの定義が
真に同時だとするなら、bを評価する際には aの値として 5を使う
べきだと⾔う。よって、Evaの⾒⽅では、aは 5となり、bは 15と
なり、結果は 20となるべきだということになる。これらの⾒⽅の
中で、(⽀持できるものがあるとすれば)あなたはどの⾒⽅を⽀持
するか。また、Evaの⾔うようにふるまう内部定義を実装する⽅
法を考えよ。26

26mit Schemeの実装者たちは、次のような理由から Alyssaを⽀持しています。まず、
Evaは原理上は正しいことになります。定義は同時だと⾒なされるべきだからです。し

練習問題 4.20: 内部定義は逐次的に⾒えるのに実際は同時だとい
うことから、内部定義を完全に回避し、代わりに特殊形式 letrec
を使うという⼈もいる。letrecは letに似ているので、そこで束
縛される変数が同時に束縛され、お互いに同じスコープを持つこ
とは不⾃然ではない。上記の例の⼿続き fは、内部⼿続きを使わ
ずに、まったく同じ意味を持つものを書くことができる。
(define (f x)

(letrec
((even? (lambda (n)

(if (= n 0) true (odd? (- n 1)))))
(odd? (lambda (n)

(if (= n 0) false (even? (- n 1))))))
⟨fの本体の残り⟩))

letrec式は以下の形式を持つ。
(letrec ((⟨var1⟩ ⟨exp1⟩) . . . (⟨varn⟩ ⟨expn⟩))
⟨body⟩)

letrec式は letの亜種で、変数 ⟨vark ⟩に初期値を与える式 ⟨expk ⟩
は、すべての letrecの束縛を含む環境の中で評価される。このこ
とによって、上の例での even?と odd?の相互再帰や、以下のよう
に 10の階乗を評価する場合のように、束縛内での再帰が使えるよ
うになる。
(letrec

((fact (lambda (n)
(if (= n 1) 1 (* n (fact (- n 1)))))))

(fact 10))

a. letrecを派⽣式として実装せよ。上の本⽂や練習問題 4.18で
⽰したように、letrec式を let式に変形すること。つまり、
letrecの変数を letによって作り、値を set!で代⼊すると
いうことになる。

かし、Evaが求めることを⾏う⼀般的で効率的な仕組みを実装するのは難しいようです。
そのような仕組みが作れない以上は、同時定義の難しいケースではエラーを発⽣させる
(Alyssaの考え)ほうが、正しくない答えを出す (Benのやり⽅)よりもよいということで
す。

b. Louis Reasonerは、内部定義について考えるうちに混乱して
しまった。⼿続きの中で defineを使いたくないなら letを
使えばいいじゃないかというのが彼の考えだ。この練習問題
で定義した fについて、式 (f 5)の評価中に ⟨fの本体の残
り⟩が評価される環境を⽰した環境図を描き、彼の考えの⽳
を⽰せ。同じ式の評価で、fの定義中の letrecの代わりに
letが使われている場合の環境図も書け。

練習問題 4.21: 驚くべきことに、練習問題 4.20での Louisの直感
は正しい。letrecを (それどころか、defineさえ)使わずに再帰
⼿続きを規定することは実際に可能だ。しかし、これを実現する⽅
法は Louisの想像よりもずっと難しいものになる。次の式は、再帰
的な階乗⼿続きを適⽤することによって 10の階乗を求めている。
27

((lambda (n)
((lambda (fact) (fact fact n))
(lambda (ft k) (if (= k 1) 1 (* k (ft ft (- k 1)))))))

10)

a. これが実際に階乗を計算することを (式を評価して)確認せ
よ。フィボナッチ数を計算する同様の式を考えよ。

b. 相互再帰の内部定義を含む以下の⼿続きについて考える。
(define (f x)

(define (even? n)
(if (= n 0) true (odd? (- n 1))))

(define (odd? n)
(if (= n 0) false (even? (- n 1))))

(even? x))

以下の空欄に式を埋め、内部定義も letrecも使わない fの
別の定義を完成させよ。

27この例は、defineを使わずに再帰⼿続きを定式化するプログラミングトリックを⽰
したものです。この種のトリックとして最も⼀般的なものは Y コンビネータ (operator)
です。これを使うと、“純粋ラムダ計算”による再帰の実装が可能です (λ計算の詳細に
ついてはStoy 1977を、また Scheme による Y コンビネータの解説についてはGabriel
1988を参照)。

(define (f x)
((lambda (even? odd?) (even? even? odd? x))
(lambda (ev? od? n)

(if (= n 0) true (od? ⟨??⟩ ⟨??⟩ ⟨??⟩)))
(lambda (ev? od? n)

(if (= n 0) false (ev? ⟨??⟩ ⟨??⟩ ⟨??⟩)))))

4.1.7 構⽂解析を実⾏から分離する
上で実装した評価器は単純なのですが、とても⾮効率的です。これは、式の構
⽂解析と式の実⾏が交互に⾏われることによります。このため、プログラムが
何回も実⾏される場合、構⽂解析が何回も⾏われてしまいます。例えば、次
の factorialの定義を使って (factorial 4)を評価することを考えてみまし
ょう。
(define (factorial n)

(if (= n 1) 1 (* (factorial (- n 1)) n)))

factorialが呼ばれるたびに、評価器は本体が if式であることを判断し、述語
を取り出す必要があります。述語を評価して値によってディスパッチを⾏うの
はその後になります。式 (* (factorial (- n 1)) n)を評価したり、部分式
(factorial (- n 1))や (- n 1)を評価したりするたびに、評価器は evalの
中で場合分けを⾏い、式が⼿続き適⽤だということを判断し、演算⼦と被演算
⼦の抽出をする必要があります。この解析は⾼コストなので、何度も⾏うのは
無駄です。
評価器を変形することで、構⽂解析が⼀度だけ実⾏されるようにしてずっ

と効率よくできます。28 まず、式と環境を引数に取る evalを⼆つに分割しま
す。⼿続き analyzeは式だけを引数に取り、構⽂解析を⾏って新しい⼿続きを
返します。この新しい⼿続きは 実⾏⼿続き (execution procedure)で、解析済み
の式を実⾏するうえでするべき仕事をカプセル化したものです。実⾏⼿続きは
環境を引数に取り、評価を完了します。こうすると、実⾏⼿続きが何度も呼ば
れたとしても、analyzeはひとつの式に対して⼀回しか呼ばれないので、作業
量を減らすことができます。

28このテクニックは、第 5章で検討するコンパイル処理には不可⽋のものです。Jonathan
Reesは、1982年ごろ Tプロジェクトのためにこのような Schemeインタプリタを書い
ています (Rees and Adams 1982)。Marc Feeley (1986) (Feeley and Lapalme 1987も参
照)は、このテクニックを修⼠論⽂で独⽴に開発しました。

解析と実⾏を分けることで、evalは以下のようになります。
(define (eval exp env) ((analyze exp) env))

analyze呼び出しの返り値は実⾏⼿続きで、それを環境に適⽤します。analyze
⼿続きは、4.1.1節の元の evalで実⾏したのと同じ場合分けです。違うのは、
ディスパッチ先の⼿続きで⾏うのが解析だけで、完全な評価を⾏わないという
ところです。
(define (analyze exp)

(cond ((self-evaluating? exp) (analyze-self-evaluating exp))
((quoted? exp) (analyze-quoted exp))
((variable? exp) (analyze-variable exp))
((assignment? exp) (analyze-assignment exp))
((definition? exp) (analyze-definition exp))
((if? exp) (analyze-if exp))
((lambda? exp) (analyze-lambda exp))
((begin? exp) (analyze-sequence (begin-actions exp)))
((cond? exp) (analyze (cond->if exp)))
((application? exp) (analyze-application exp))
(else (error "Unknown expression type: ANALYZE" exp))))

⾃⼰評価式を扱う最も単純な形の構⽂解析は以下のようになります。環境引数
を無視して式を返すだけの実⾏⼿続きを返します。
(define (analyze-self-evaluating exp)

(lambda (env) exp))

クォート式については、実⾏フェーズではなく解析フェーズでテキストを⼀度
だけ取り出すことになるので、少しだけ効率をよくすることが可能です。
(define (analyze-quoted exp)

(let ((qval (text-of-quotation exp)))
(lambda (env) qval)))

変数の値の探索は、依然として実⾏フェーズで⾏う必要があります。変数の値
の検索は環境の知識に依存するからです。29

29しかし、変数探索の中にも、構⽂解析の⼀部として実⾏可能な重要な部分がありま
す。5.5.6節で⽰すように、環境構造の中で変数の値が⾒つかる場所は決定できます。そ
のため、変数にマッチするエントリを探すために環境を⾛査しないでもよくなります。

(define (analyze-variable exp)
(lambda (env) (lookup-variable-value exp env)))

analyze-assignmentも、環境が与えられる実⾏時まで、変数を実際に設定す
るのを遅らせる必要があります。しかし、解析時に assignment-value式を (再
帰的に)解析できるというのは効率の向上に⼤きく寄与します。assignment-
value式は⼀回しか解析されないことになるからです。定義についても同じこ
とが⾔えます。
(define (analyze-assignment exp)

(let ((var (assignment-variable exp))
(vproc (analyze (assignment-value exp))))

(lambda (env)
(set-variable-value! var (vproc env) env)
'ok)))

(define (analyze-definition exp)
(let ((var (definition-variable exp))

(vproc (analyze (definition-value exp))))
(lambda (env)

(define-variable! var (vproc env) env)
'ok)))

if式については、解析時に述語、結果式、代替式を取り出し、解析します。
(define (analyze-if exp)

(let ((pproc (analyze (if-predicate exp)))
(cproc (analyze (if-consequent exp)))
(aproc (analyze (if-alternative exp))))

(lambda (env) (if (true? (pproc env))
(cproc env)
(aproc env)))))

lambda式の解析も、⼤きな効率向上が可能です。lambdaを評価した結果の⼿
続きが何回も適⽤されることになるとしても、lambdaの本体は⼀回しか解析
されないからです。
(define (analyze-lambda exp)

(let ((vars (lambda-parameters exp))
(bproc (analyze-sequence (lambda-body exp))))

(lambda (env) (make-procedure vars bproc env))))

式の列 (beginや、lambda式の本体に出てくるようなもの)の解析はより複雑
です。30 列のそれぞれの式は解析され、実⾏⼿続きが作られます。これらの実
⾏⼿続きは、引数に環境を取り、その環境を引数としてそれぞれの実⾏⼿続き
を順番に呼び出していくひとつの実⾏⼿続きにまとめられます。
(define (analyze-sequence exps)

(define (sequentially proc1 proc2)
(lambda (env) (proc1 env) (proc2 env)))

(define (loop first-proc rest-procs)
(if (null? rest-procs)

first-proc
(loop (sequentially first-proc (car rest-procs))

(cdr rest-procs))))
(let ((procs (map analyze exps)))

(if (null? procs) (error "Empty sequence: ANALYZE"))
(loop (car procs) (cdr procs))))

適⽤を解析するには、まず演算⼦と被演算⼦を解析します。それから、(実際に
適⽤する⼿続きを得るために)演算⼦の実⾏⼿続きを呼び出し、(実際の引数を
得るために) 被演算⼦の実⾏⼿続きを呼び出す実⾏⼿続きを構築します。次に、
これらを execute-applicationに渡します。これは4.1.1節の applyと似たよ
うなものです。execute-applicationが applyと違うのは、複合⼿続きの本体
は解析済みで、それ以上の解析を⾏う必要がないということです。代わりに⾏
うのは、拡張した環境を引数として本体の実⾏⼿続きを呼び出すことだけです。
(define (analyze-application exp)

(let ((fproc (analyze (operator exp)))
(aprocs (map analyze (operands exp))))

(lambda (env)
(execute-application
(fproc env)
(map (lambda (aproc) (aproc env))

aprocs)))))
(define (execute-application proc args)

(cond ((primitive-procedure? proc)
(apply-primitive-procedure proc args))

30列の処理について理解を深めるには練習問題 4.23を参照してください。

((compound-procedure? proc)
((procedure-body proc)
(extend-environment
(procedure-parameters proc)
args
(procedure-environment proc))))

(else
(error "Unknown procedure type: EXECUTE-APPLICATION"

proc))))

新しい評価器は、4.1.2節, 4.1.3節, 4.1.4節に出てきたものと同じデータ構造、
構⽂⼿続き、実⾏時サポート⼿続きを使います。

練習問題 4.22: この節の評価器を拡張し、特殊形式 letをサポー
トするようにせよ (練習問題 4.6参照)。
練習問題 4.23: Alyssa P. Hackerは、analyze-sequenceがどうし
てこんなに複雑になるのか理解できない。ほかの解析⼿続きは、ど
れも4.1.1節に出てきた対応する評価⼿続き (または eval節)を素
直に変形したものだ。彼⼥は、analyze-sequenceは以下のように
なると思っていた。
(define (analyze-sequence exps)

(define (execute-sequence procs env)
(cond ((null? (cdr procs))

((car procs) env))
(else
((car procs) env)
(execute-sequence (cdr procs) env))))

(let ((procs (map analyze exps)))
(if (null? procs)

(error "Empty sequence: ANALYZE"))
(lambda (env)

(execute-sequence procs env))))

Eva Lu Atorは、本⽂中のバージョンでは解析時に列を評価する
という仕事以上のことをやっていると Alyssaに説明する。Alyssa
の列実⾏⼿続きは、個別の実⾏⼿続きに対する呼び出しを組み込
むのではなく、⼿続きを呼ぶためにループしている。実質的には、

列内の個別の式は解析されているが、列そのものは解析されてい
ないことになる。
analyze-sequenceの両バージョンを⽐較せよ。例えば、列が式を
ひとつだけ含んでいるというよくあるケース (⼿続きの本体では
それが普通だ)について考える。Alyssaのプログラムに⽣成され
た実⾏⼿続きはどのような仕事をするか。上の本⽂中のプログラ
ムに⽣成された実⾏⼿続きについてはどうだろうか。⼆つの式を
持つ列については、両バージョンはどのように異なるだろうか。
練習問題 4.24: 元のメタ循環評価器とこの節のバージョンの速度
を⽐較する実験をいくつか設計し、実⾏せよ。その結果を使って、
いろいろな⼿続きについて、解析と実⾏に使われる時間の割合を
⾒積もれ。

4.2 Schemeの変異版 — 遅延評価
Lispのプログラムとして表現された評価器を⼿に⼊れたので、評価器を修正
するだけで⾔語設計上の別の選択肢の実験を⾏うことができるようになりまし
た。実際、新⾔語の開発は、最初は既存の⾼級⾔語の中に新しい⾔語を埋め込
む評価器を書くことによって⾏われることが多いのです。例えば、Lispの修
正提案のいくつかの側⾯について Lispコミュニティのほかのメンバと議論を
したければ、その変更を実装した評価器を提供できます。こうすると、受け取
った⼈はその新しい評価器で実験を⾏い、さらなる修正という形でコメントを
返すことができます。⾼⽔準の実装基盤は、評価器のテストとデバッグを簡単
にするだけではありません。既存の⾔語に埋め込むことで、ここでの組み込み
Lisp評価器が基層の Lispの基本要素や制御構造を使っているのと同じように、
⾔語を設計する⼈が機能をパクる (snarf) 31 ことができるようになります。⾔
語設計者が低レベル⾔語やハードウェアで完全な実装を (必要に応じて)構築
するのは、後になってからでも⼤丈夫です。この節と次の節では、表現⼒を⼤
きく向上させる Schemeの変異版について探求していきます。

31Snarf: “(特に、⼤きな⽂書やファイルを)所有者の許可を得て、または許可を得ずに、
使うために取ってくること。” Snarf down: “snarfすること。時に、吸収する・処理する
・理解するという意味を帯びる。”(これらの定義はSteele et al. 1983から snarfしたもの
です。Raymond 1993も参照。)

4.2.1 正規順序と適⽤順序
1.1節で評価モデルの議論を始めたとき、Schemeは適⽤順序 (applicative-order)
⾔語だと述べました。つまり、Schemeの⼿続きの引数はすべて⼿続きの適⽤
時に評価されるということです。それに対して、正規順序 (normal-order)⾔語
では、実際の引数の値が必要になるまで⼿続き引数の評価を遅らせます。⼿続
き引数の評価を可能な限り (例えば、基本演算に必要になるまで)遅らせること
は、遅延評価 (lazy evaluation)と呼ばれます。32 以下の⼿続きについて考えて
みましょう。
(define (try a b) (if (= a 0) 1 b))

(try 0 (/ 1 0))を評価すると、Schemeではエラーになります。遅延評価の
場合はエラーは起こりません。この式を評価すると 1を返します。引数の (/ 1
0)は⼀度も評価されることがないからです。
次の⼿続き unlessの定義は、遅延評価を利⽤した例となっています。

(define (unless condition usual-value exceptional-value)
(if condition exceptional-value usual-value))

これは、次のような式で使うことができます。
(unless (= b 0)

(/ a b)
(begin (display "exception: returning 0") 0))

これは、適⽤順序の⾔語では動作しません。unlessが呼ばれる前に、通常の値
と例外の値が両⽅とも評価されてしまうためです (練習問題 1.6と⽐較してく
ださい)。遅延評価の利点のひとつは、unlessのようなある種の⼿続きが、引
数のうち評価するとエラーになったり停⽌しなかったりするものがあったとし
ても、役に⽴つ計算ができるということです。
ある引数を評価し終わる前に⼿続きの本体に⼊るという場合、その⼿続き

はその引数について ⾮厳密 (non-strict)であると⾔います。ある引数が⼿続き
の本体に⼊る前に評価される場合、その⼿続きはその引数について厳密 (strict)

32“遅延”という⽤語と “正規順序”という⽤語の違いは少々曖昧です。⼀般的に、“遅
延”は特定の評価器の仕組みについて⾔いますが、“正規順序”は特定の評価戦略とは独
⽴の⾔語の意味論について⾔います。しかし、これは厳格な区別というわけではなく、⼆
つの⽤語を区別せずに使われることもよくあります。

であると⾔います。33純粋な適⽤順序⾔語では、すべての複合⼿続きはすべて
の引数について厳密で、基本⼿続きは厳密であることも⾮厳密であることもあ
ります。このほかに、定義する⼿続きの厳密さをプログラマが細かくコントロ
ールできるようにしている⾔語もあります (練習問題 4.31参照)。
⾮厳密にすることで有⽤な結果が得られる⼿続きの顕著な例は cons(に限

らず、⼀般的にほとんどすべてのデータ構造のコンストラクタ)です。要素の値
がわかっていない場合でも、要素を組み合わせ、データ構造を作り、結果とな
るデータ構造に対して演算を⾏うことで、有⽤な計算ができます。例えば、リ
ストの個別の要素の値を知らないままでリストの⻑さを計算するという考え⽅
には、どこにもおかしなところはありません。4.2.3節では、この考えを利⽤し、
⾮厳密な consペアからなるリストとして第 3章のストリームを実装します。

練習問題 4.25: unlessを (通常の適⽤順序 Schemeで)上で⽰した
ように定義し、それから factorialを unlessを使って以下のよ
うに定義するとする。
(define (factorial n)

(unless (= n 1)
(* n (factorial (- n 1)))
1))

(factorial 5)を評価すると何が起こるだろうか。この定義は正
規順序⾔語でも動くだろうか。
練習問題 4.26: Ben Bitdiddleと Alyssa P. Hackerは、unlessの
ようなものを実装する際に遅延評価がどれだけ重要かというとこ
ろで意⾒が分かれている。Benは、適⽤順序でも特殊形式として
unlessを実装できると指摘した。Alyssaは、そういうふうにして
しまうと、unlessはただの構⽂になってしまって、⾼階⼿続きと
組み合わせて使える⼿続きにはならないと反論した。両者の議論
の細部を補え。unlessを (condや letのように)派⽣⼿続きとし
て実装する⽅法を⽰せ。また、unlessを特殊形式でなく⼿続きと
して持つほうが役に⽴つ状況の例を挙げよ。

33“厳密”、“⾮厳密”という⽤語は、本質的には “適⽤順序”、“正規順序”と同じ意味で
す。違うのは、“厳密”、“⾮厳密”は個別の⼿続きと引数について⾔うもので、⾔語全体
について⾔うものではないというところです。プログラミング⾔語のカンファレンスで
は、次のような話を聞くかもしれません。“正規順序⾔語の Hassleは、厳密な基本⼿続
きをいくつか持っています。そのほかの⼿続きは引数を遅延評価で取ります。”

436



4.2.2 遅延評価のインタプリタ
この節では、正規順序の⾔語を実装していきます。この⾔語は Schemeと同じ
ですが、複合⼿続きがすべての引数について⾮厳密であるという点が違います。
基本⼿続きは厳密なままです。4.1.1節の評価器を修正して、解釈対象の⾔語が
そのようにふるまうようにするのは難しくありません。必要な変更は、ほとん
どすべて⼿続き適⽤の付近に固まっています。
ここでの基本的な考え⽅は、⼿続き適⽤の際に、どの引数を評価してどの

引数を遅延させるかインタプリタが決めないといけないということです。遅延
された引数は評価されません。その代わり、それらの引数はサンク (thunk)訳注 5

と呼ばれるオブジェクトに変形されます。34サンクは、引数の値が必要になっ
たときに、あたかも適⽤時に評価されていたかのようにその値を返します。サ
ンクはそのために必要な情報を含んでいる必要があります。よって、サンクは
引数の式と、⼿続き適⽤が評価された環境を持っていなければならないことに
なります。
サンク中の式を評価するプロセスは強制 (forcing)と呼ばれます。35 ⼀般的

に、サンクが強制されるのはその値が必要になったときだけです。これには、
サンクの値を使う基本⼿続きにサンクが渡される場合、条件式の述語の値であ
る場合、サンクが⼿続きとして適⽤される演算⼦の値である場合があります。
設計上の選択肢として、3.5.1節で遅延オブジェクトに対して⾏ったように、サ
ンクをメモ化 (memoize)するかどうかということがあります。メモ化をする場
合、サンクが初めて強制されたとき、サンクは計算された値を格納します。そ
れ以降に⾏われる強制では、計算を再び⾏うことはせずに、格納された値を単
純に返します。ここでは、インタプリタがメモ化を⾏うようにします。多くの
応⽤ではそのほうが効率的だからです。しかし、これには気をつけないといけ
ないところもあります。36

訳注 5“think”の過去分詞は本来は “thought” ですが、考案者たちが夜中まで議論していて
頭が朦朧としているときにつけたのでこの形になったという説があります。

34サンク (thunk)という単語は、Algol 60の名前呼びの実装について議論していた⾮公
式な作業部会によって考え出されたものです。彼らは、式のほとんどの分析 (“thinking
about”)はコンパイル時にできるということに気づきました。つまり、実⾏時には式はす
でに「考えられて」(“thunk” about(Ingerman et al. 1960))いることになります。

35これは、第 3章でストリームを表現するために導⼊した遅延オブジェクトに対する
forceの使⽤と似ています。ここで⾏っていることと第 3章で⾏ったことの決定的な違
いは、ここでは遅延と強制を評価器に組み込んでいて、⾔語全体を通して統⼀的かつ⾃
動的に扱えるようにしているというところです。

36名前呼び (call-by-name) 引数渡しに対して、メモ化と遅延評価を組み合わせたもの

評価器の修正
遅延評価器と4.1節の評価器との主な違いは、evalと applyでの⼿続き適⽤の
扱いです。

evalの application?節は次のようになります。
((application? exp)
(apply (actual-value (operator exp) env)

(operands exp)
env))

これは4.1.1節の evalの application?節とほとんど同じです。しかし、遅延
評価では、applyの引数として、被演算⼦式を評価して得られたものを渡すの
ではなく、被演算⼦式をそのまま渡します。引数を遅延化する場合にはサンク
の構築に環境が必要になるので、環境も⼀緒に渡す必要があります。applyが
⼿続きの型 (基本⼿続きもしくは複合⼿続き)によってディスパッチを⾏い⼿
続きを適⽤するためには実際の⼿続きが必要となるため、演算⼦に対しては遅
延評価器も評価を⾏います。
式の実際の値が必要になったときには、単純に evalをする代わりに以下の

ようにします。
(define (actual-value exp env)

(force-it (eval exp env)))

こうすることで、式の値がサンクであれば強制されることになります。
applyの新しいバージョンも、4.1.1節のバージョンとほとんど同じです。

違うのは、evalが未評価の被演算⼦式を渡してくるというところです。基本⼿
続きについては、(厳密であるため)適⽤する前にすべての引数を評価します。
複合⼿続きについては、(⾮厳密であるため) ⼿続きに適⽤する前に遅延化し
ます。
(define (apply procedure arguments env)

(cond ((primitive-procedure? procedure)

は、必要呼び (call-by-need)引数渡しと呼ばれることがあります。(call-by-nameはAlgol
60で導⼊されたものですが、これはメモ化しない遅延評価に似ています)。⾔語設計者で
ある私たちは、評価器にメモ化させることも、メモ化させないことも、その選択をプロ
グラマに任せることもできます (練習問題 4.31)。第 3章で⾒てきたことから予測できる
ように、代⼊がある場合、この選択は難しく混乱を招く問題を引き起こします (練習問題
4.27、練習問題 4.29参照)。Clinger (1982)による優れた論⽂では、ここで起こる混乱の
複数の側⾯について明らかにすることを試みています。

(apply-primitive-procedure
procedure
(list-of-arg-values arguments env))) ; changed

((compound-procedure? procedure)
(eval-sequence
(procedure-body procedure)
(extend-environment
(procedure-parameters procedure)
(list-of-delayed-args arguments env) ; changed
(procedure-environment procedure))))

(else (error "Unknown procedure type: APPLY"
procedure))))

引数を処理する⼿続きはほとんど4.1.1節の list-of-valuesそのままですが、
違うのは list-of-delayed-argsが引数を評価するのではなく遅延化すること、
list-of-arg-valuesが evalではなく actual-valueを使うということです。
(define (list-of-arg-values exps env)

(if (no-operands? exps)
'()
(cons (actual-value (first-operand exps)

env)
(list-of-arg-values (rest-operands exps)

env))))
(define (list-of-delayed-args exps env)

(if (no-operands? exps)
'()
(cons (delay-it (first-operand exps)

env)
(list-of-delayed-args (rest-operands exps)

env))))

ほかに評価器を修正しなければならないところは if の処理の中にあります。
ここでは、述語式の真偽をテストする前に、evalではなく actual-valueを使
って述語式の値を取る必要があります。
(define (eval-if exp env)

(if (true? (actual-value (if-predicate exp) env))

(eval (if-consequent exp) env)
(eval (if-alternative exp) env)))

最後に、4.1.4節の driver-loop⼿続きを変更して、evalの代わりに actual-
valueを使うようにする必要があります。これは、遅延化された値が REPLま
で返された場合、表⽰前に強制するようにするためです。また、この評価器が
遅延評価器であることを⽰すためにプロンプトも変更しておきます。
(define input-prompt ";;; L-Eval input:")
(define output-prompt ";;; L-Eval value:")
(define (driver-loop)

(prompt-for-input input-prompt)
(let ((input (read)))

(let ((output
(actual-value
input the-global-environment)))

(announce-output output-prompt)
(user-print output)))

(driver-loop))

これらの変更を⾏ったら、評価器を動かしてテストできます。4.2.1節で検討し
た try式の評価がうまくいったら、インタプリタが遅延評価を実⾏していると
いうことになります。
(define the-global-environment (setup-environment))
(driver-loop)
;;; L-Eval input:
(define (try a b) (if (= a 0) 1 b))
;;; L-Eval value:
ok
;;; L-Eval input:
(try 0 (/ 1 0))
;;; L-Eval value:
1

サンクの表現
この評価器は、⼿続きが引数に適⽤されるときにサンクを作り、後でそれらの
サンクを強制するようにしなければなりません。サンクは、後で引数を⽣成で

きるように、式と環境をまとめておく必要があります。サンクを強制するには、
単純にサンクから式と環境を取り出し、その環境の中で式を評価します。ここ
では evalではなく actual-valueを使います。これは、式の値がそれ⾃⾝サン
クであった場合、サンクでないものにたどり着くまで強制を続けるようにする
ためです。
(define (force-it obj)

(if (thunk? obj)
(actual-value (thunk-exp obj) (thunk-env obj))
obj))

式と環境をまとめる簡単な⽅法のひとつは、式と環境を含むリストを作るとい
うことです。そこで、サンクは次のように作ることにします。
(define (delay-it exp env)

(list 'thunk exp env))
(define (thunk? obj)

(tagged-list? obj 'thunk))
(define (thunk-exp thunk) (cadr thunk))
(define (thunk-env thunk) (caddr thunk))

実際のところは、評価器に⼊れたいものはこれとはちょっと違います。必要な
のはメモ化されたサンクです。そのため、サンクが強制されたときには格納さ
れた式をその値で置き換え、thunkタグを別のものに変えて、サンクが評価済
みだということがわかるようにします。37

(define (evaluated-thunk? obj)
(tagged-list? obj 'evaluated-thunk))

(define (thunk-value evaluated-thunk)
(cadr evaluated-thunk))

37式の値を評価したときに、ついでにサンクから envを消しています。こうしても、イ
ンタプリタから返される値は変わりません。しかし、空間の節約にはなります。5.3節で
検討するように、必要がなくなり次第サンクから envへの参照を削除するようにすると、
この構造はガベージコレクション (garbage-collect)され、空間はリサイクルされるから
です。
同様に、3.5.1節のメモ化した遅延オブジェクトの場合も、必要のなくなった環境は削

除できるところです。そのためには、memo-procで (set! proc '())のようなことを⾏
って、(delayが評価された環境を含む)⼿続き procを値を格納した後に捨てるようにし
ます。

(define (force-it obj)
(cond ((thunk? obj)

(let ((result (actual-value (thunk-exp obj)
(thunk-env obj))))

(set-car! obj 'evaluated-thunk)
(set-car! (cdr obj)

result) ; expをその値で置き換える
(set-cdr! (cdr obj)

'()) ; 必要のなくなった envを忘れる
result))

((evaluated-thunk? obj) (thunk-value obj))
(else obj)))

delay-it⼿続きは、そのままでメモ化ありでもメモ化なしでも動くというと
ころに注意してください。

練習問題 4.27: 以下の定義を遅延評価器に⼊⼒したとする。
(define count 0)
(define (id x) (set! count (+ count 1)) x)

以下の⼀連のやりとりの空欄の値を埋め、回答について説明せよ。
38

(define w (id (id 10)))
;;; L-Eval input:
count
;;; L-Eval value:
⟨response⟩
;;; L-Eval input:
w
;;; L-Eval value:
⟨response⟩
;;; L-Eval input:
count
;;; L-Eval value:
⟨response⟩

38この練習問題は、遅延評価と副作⽤との⼲渉や副作⽤が⼤きな混乱のもとになると
いうことを⽰しています。これは、第 3章の議論から予想できる範囲内のことです。

練習問題 4.28: evalは、演算⼦を applyに渡す前に、evalではな
く actual-valueを使って演算⼦を評価する。これは、演算⼦の値
を強制するためである。ここで強制を⾏う必要があるということ
がわかるような例を⽰せ。
練習問題 4.29: メモ化しなければ、メモ化した場合よりも実⾏が
ずっと遅くなりそうなプログラムを⽰せ。また、以下の対話につい
て考えよ。この対話の中では、id⼿続きの定義は 練習問題 4.27と
同じで、countは 0から始める。
(define (square x) (* x x))
;;; L-Eval input:
(square (id 10))
;;; L-Eval value:
⟨response⟩
;;; L-Eval input:
count
;;; L-Eval value:
⟨response⟩

評価器がメモ化されている場合とメモ化されていない場合の両⽅
について、応答を⽰せ。
練習問題 4.30: 改⼼した C⾔語プログラマの Cy D. Fectは、遅延
評価器では列の中の式を強制しないので、副作⽤が起こらないこ
とになるのではないかと⼼配している。列の中の式の値は最後の
もの以外は使われない (式は変数代⼊や表⽰などの作⽤のために存
在する)ので、(例えば基本⼿続きの引数として) 値が使われて強制
されるということが起こらないかもしれないと⾔うのだ。そのた
め Cyは、列を評価するときは、最後の式以外のすべての式を強制
するべきだと考えている。彼は、4.1.1節の eval-sequenceを修正
し、evalでなく actual-valueを使⽤したほうがよいと提案した。
(define (eval-sequence exps env)

(cond ((last-exp? exps) (eval (first-exp exps) env))
(else (actual-value (first-exp exps) env)

(eval-sequence (rest-exps exps) env))))

a. Ben Bitdiddleは、Cyは間違っていると考えている。彼は、練
習問題 2.23で説明した for-each⼿続きを Cyに⾒せた。こ
れは、副作⽤のある列の重要な例となっている。
(define (for-each proc items)

(if (null? items)
'done
(begin (proc (car items))

(for-each proc (cdr items)))))

彼は、本⽂中の評価器 (元の eval-sequenceを持つ)はこれ
を正しく扱えると主張する。
;;; L-Eval input:
(for-each (lambda (x) (newline) (display x))

(list 57 321 88))
57
321
88
;;; L-Eval value:
done

for-eachのふるまいについて、なぜ Benが正しいのか説明
せよ。

b. Cyは、for-eachについては Benが正しいと認めたが、eval-
sequenceに対する変更を提案したときに考えていたのはこ
の種のプログラムではないと⾔う。彼は、次の⼆つの⼿続き
を遅延評価器で定義した。
(define (p1 x)

(set! x (cons x '(2)))
x)

(define (p2 x)
(define (p e)

e
x)

(p (set! x (cons x '(2)))))

元の eval-sequenceを使うと、(p1 1)と (p2 1)の値はどう
なるだろうか。Cyが提案する eval-sequenceに対する変更
を適⽤した場合はどうだろうか。

c. Cyはまた、彼の提案する eval-sequenceに対する変更は、a
の例でのふるまいに影響を与えないと指摘した。なぜこれが
正しいのか説明せよ。

d. あなたの考えでは、遅延評価器で列を扱うのはどのようにす
るのがよいだろうか。Cyのアプローチだろうか、本⽂中の
アプローチだろうか。それとも、別の何かのアプローチだろ
うか。

練習問題 4.31: この節でのアプローチはちょっと気持ちの悪いと
ころがある。というのは、互換性のない変更を Schemeに加えて
しまっているからだ。遅延評価を上位互換性のある拡張 (upward-
compatible extension)として、つまり通常の Schemeが前と同じよ
うに動くように実装できれば、そのほうがよいだろう。これは、⼿
続き定義の構⽂を拡張し、引数を遅延させるかどうかをユーザー
がコントロールできるようにすれば実現できる。ついでに、遅延
をメモ化するかどうかの選択もユーザーができるようにしてもよ
いだろう。例えば、
(define (f a (b lazy) c (d lazy-memo))
. . .)

この定義は、4引数の⼿続きで、1番⽬と 3番⽬の引数は⼿続きが
呼ばれたときに評価され、2番⽬の引数は遅延化され、4番⽬の引
数は遅延化とメモ化が⾏われるものとして fを定義する。つまり、
通常の⼿続き定義をすると通常の Schemeと同じふるまいになる
が、すべての複合⼿続きの各仮引数に lazy-memo宣⾔を加えると、
この節で定義した遅延評価器のふるまいになる。Schemeにそのよ
うな拡張を加えるのに必要な変更を設計し、実装せよ。defineの
新しい構⽂を扱うには、新しい構⽂⼿続きを実装する必要がある。
また、evalや applyが引数をどういうタイミングで遅延化するか
を決定できるようにし、その決定に従って引数を強制したり遅延
化したりできるようにしなければならない。強制についてメモ化
するかどうかも適切に扱えるようにする必要がある。

4.2.3 遅延化リストとしてのストリーム
3.5.1節では、遅延化されたリストとしてストリームを実装するやり⽅を⽰し
ました。そのときは、特殊形式 delayと cons-streamを導⼊し、ストリームの
cdrを計算するという “promise”(約束)を、実際の実⾏を先延ばしにしながら
構築できるようにしました。評価プロセスについてコントロールしたいことが
増えるたびに、この特殊形式を導⼊するという⼀般的なやり⽅を使うことはで
きますが、不格好です。問題のひとつとして、特殊形式が⼿続きのような第⼀
級オブジェクトではないので、⾼階⼿続きと組み合わせて使えないということ
があります。39 そのうえ、リストに似ているもののリストと同⼀ではない新し
い種類のデータオブジェクトとしてストリームを作らざるを得ず、そのことに
よって通常のリスト演算の多く (map, append など) をストリーム⽤に再実装し
なければならないことになりました。
遅延評価があれば、ストリームとリストは同じものにできます。そうする

と、特殊形式を作ったり、リストとストリームに別々の演算を⽤意したりする
必要はなくなります。そのためには、consが⾮厳密になるようにできれば⼗
分です。これを実現するひとつの⽅法は、遅延評価器を拡張して⾮厳密な基本
⼿続きを許すようにして、そのひとつとして consを実装するということです。
もっと簡単な⽅法は、consを基本⼿続きとして実装する根本的な必要性はな
い (2.1.3節) ということを思い出して、ペアを⼿続きとして表現するというこ
とです。40

(define (cons x y) (lambda (m) (m x y)))
(define (car z) (z (lambda (p q) p)))
(define (cdr z) (z (lambda (p q) q)))

これらの基本演算を使うと、リスト演算の標準定義は有限リストだけでなく無
限リスト (ストリーム)に対しても動作するようになり、ストリーム演算はリス
ト演算として実装できることになります。以下にいくつかの例を⽰します。
(define (list-ref items n)

(if (= n 0)

39これはまさしく、練習問題 4.26の unlessで起こった問題です。
40この⼿続きによる表現は練習問題 2.4で説明したものです。本質的には、どんな⼿続
きによる表現 (例えば、メッセージパッシングによる実装)を使っても問題はありません。
これらの定義は、ドライバループに⼊⼒すれば遅延評価器に組み込むことができます。元
々 cons, car, cdrをグローバル環境の基本⼿続きに⼊れていたとしたら、それらは再定
義されることになります (練習問題 4.33と練習問題 4.34も参照してください)。

(car items)
(list-ref (cdr items) (- n 1))))

(define (map proc items)
(if (null? items)

'()
(cons (proc (car items)) (map proc (cdr items)))))

(define (scale-list items factor)
(map (lambda (x) (* x factor)) items))

(define (add-lists list1 list2)
(cond ((null? list1) list2)

((null? list2) list1)
(else (cons (+ (car list1) (car list2))

(add-lists (cdr list1) (cdr list2))))))
(define ones (cons 1 ones))
(define integers (cons 1 (add-lists ones integers)))
;;; L-Eval input:
(list-ref integers 17)
;;; L-Eval value:
18

これらの遅延リストは、第 3章のストリームよりもさらに遅延化されていると
いうことに注意してください。この場合、リストの cdrだけでなく、carも同
様に遅延化されています。41 実際のところ、遅延ペアの carや cdrにアクセス
する場合であっても、リスト要素の値を強制する必要はありません。値は、本
当に必要になったときになって—例えば、基本⼿続きの引数になる場合や答え
として表⽰される場合など—初めて強制されることになります。
遅延ペアは、3.5.4節でストリームを扱った際に起こったような問題を解決

するのにも役に⽴ちます。そのときの問題は、ループのあるシステムのストリ
ームモデルを定式化するには、cons-streamによって提供されるもの以外にも、
delayを明⽰的にプログラムの中にちりばめないといけないということでした。
遅延評価では、⼿続きのすべての引数は統⼀的に遅延化されます。例えば、リ
ストを積分して微分⽅程式を解く⼿続きを、3.5.4節で元々やりたかったよう
に実装できます。
(define (integral integrand initial-value dt)

41このことによって、列に限らず、より⼀般的なリスト構造の遅延化版を作れるように
なっています。Hughes 1990では、“遅延⽊”のいくつかの応⽤について検討しています。

(define int
(cons initial-value

(add-lists (scale-list integrand dt) int)))
int)

(define (solve f y0 dt)
(define y (integral dy y0 dt))
(define dy (map f y))
y)

;;; L-Eval input:
(list-ref (solve (lambda (x) x) 1 0.001) 1000)
;;; L-Eval value:
2.716924

練習問題 4.32: 第 3章のストリームと、この節で説明した “より遅
延化された”遅延リストとの間の違いを⽰す例をいくつか挙げよ。
この遅延度の⾼さを利⽤するにはどのようにすればいいだろうか。
練習問題 4.33: Ben Bitdiddleは、次の式を評価することで、上記
の遅延リストの実装をテストした。
(car '(a b c))

驚いたことに、これはエラーになった。少し考えて、彼はクォート
式を読み込むことによって得られる “リスト”は cons, car, cdrの
新しい定義によって操作されるリストとは違うということに気が
ついた。評価器のクォート式の扱いを修正し、ドライバループに
⼊⼒されたクォート式のリストが真の遅延リストを⽣成するよう
にせよ。
練習問題 4.34: 評価器のドライバループを修正し、遅延ペアと遅
延リストが何らかの⽅式で適切に表⽰されるようにせよ (無限リ
ストについてはどうするのがよいだろうか？)。評価器が表⽰する
際に遅延ペアを識別できるように、遅延ペアの表現も修正する必
要があるかもしれない。

4.3 Schemeの変異版 — ⾮決定性計算
この節では、Scheme評価器を拡張して、⾮決定性計算 (nondeterministic com-
puting)と呼ばれるプログラミングパラダイムをサポートするようにします。そ
のために、⾃動探索をサポートする機能を評価器に組み込みます。これは、4.2
節で遅延評価を導⼊したときに⽐べて、⾔語に対してずっと根本的な変更を⾏
うことになります。
⾮決定性計算は、ストリーム処理と同じように、“⽣成してテストする”応

⽤に役に⽴ちます。例えば、⼆つの正の整数のリストから始めて、和が素数と
なる整数ペア (ひとつは⼀つ⽬のリストから取ったもの、ひとつは⼆つ⽬のリ
ストから取ったもの)を探すというタスクについて考えてみましょう。有限列
の演算によってこの問題を扱う⽅法については2.2.3節で、無限ストリームを
使う⽅法については3.5.3節で⾒てきました。やり⽅は、すべての可能なペアの
列を⽣成し、それらをフィルタして和が素数になるペアを選択するというもの
でした。第 2章のように実際にペアの列全体をまず⽣成するか、第 3章のよう
に⽣成とフィルタリングを交互に配置するかというところは、計算がどのよう
に構成されるかという本質的なイメージに⽐べると些細な問題です。
⾮決定性のアプローチは、違うイメージになります。単純に⼀つ⽬のリス

トから数値を (何らかの⽅法で) ひとつ選び、⼆つ⽬のリストからも数値をひ
とつ選び、それらの和が素数であるよう (何らかの仕組みによって)要求する
というものを想像してみてください。これは次の⼿続きによって表現できます。
(define (prime-sum-pair list1 list2)

(let ((a (an-element-of list1))
(b (an-element-of list2)))

(require (prime? (+ a b)))
(list a b)))

この⼿続きは、問題をただ書き直しただけで、解き⽅を規定しているようには
⾒えないかもしれません。しかしながら、これはちゃんとした⾮決定性プログ
ラムになっています。42

42ここでは、数値が素数かどうかテストする⼿続き prime?をすでに定義してあるとし
ています。prime?が定義されているとしても、prime-sum-pair⼿続きは、1.1.7節の最
初で記述した、平⽅根関数を定義しようとした役に⽴たない “Lispもどき”に怪しいぐ
らいそっくりに⾒えるかもしれません。実は、あの考え⽅による平⽅根⼿続きは、実際
に⾮決定性プログラムとして定式化できます。探索メカニズムを評価器に組み込むこと
で、純粋に宣⾔的な記述と、答えを計算する⽅法の命令的な仕様との間の区別は曖昧に

ここで鍵となる考え⽅は、⾮決定性⾔語の式は⼆つ以上の可能な値を持つ
ことができるということです。例えば、an-element-ofは与えられたリストの
どの要素を返す可能性もあります。ここでの⾮決定性プログラム評価器は、⾃
動的に可能な値を選択し、その選択を覚えておくことによって動作します。後
続の要求が満たされなければ、評価器は別の選択を試します。そして、評価が
成功するか、または選ぶものがなくなるまで、新しい選択を試し続けることに
なります。遅延評価器によってプログラマが値を遅延化したり強制したりする
⽅法の細かいところを考えないでもよくなったように、⾮決定性プログラム評
価器の場合は、プログラマは選ぶ⽅法の細かいところを考えないでもいいよう
になります。
⾮決定性評価とストリーム処理がそれぞれ連想させる、時間についての異

なるイメージを⽐べると考えを深める役に⽴ちます。ストリーム処理は、可能
な答えのストリームが組み⽴てられる時間と実際のストリームの要素が⽣成さ
れる時間とを切り離すために、遅延評価を使います。評価器は、すべての可能
な答えが時間のない列として⽬の前に並べられているという幻想を⽀えていま
す。⾮決定性評価の場合、式は可能な世界の集合の探索を表現しています。そ
れぞれの可能世界は、ある選択の集合によって決められたものです。⾮決定性
プログラム評価器は、時間は分岐していて、プログラムは複数の可能な実⾏履
歴を持っているという幻想を⽀えています。⾏き⽌まりにたどり着いた場合は、
前の選択地点に戻って異なる分岐に沿って進むことができます。
以下で実装する⾮決定性プログラム評価器は、ambと呼ばれる新しい特殊

形式に基づいているため、amb評価器と呼ばれます。上記の prime-sum-pair
の定義を (prime?, an-element-of, requireの定義とともに)amb評価器のドラ
イバループに⼊⼒すると、次のように⼿続きを実⾏できます。
;;; Amb-Eval input:
(prime-sum-pair '(1 3 5 8) '(20 35 110))
;;; Starting a new problem
;;; Amb-Eval value:
(3 20)

返り値は、条件に合う選択が⾏われるまで評価器がそれぞれのリストから要素
を選択し続けた結果として得られたものです。

4.3.1節では ambを導⼊し、どのようにしてそれが評価器の⾃動探索メカニ
ズムを通じて⾮決定性を⽀えているかを説明します。4.3.2節では⾮決定性プ
なってきます。4.4節では、この路線をさらに進めていきます。

ログラムの例を⽰し、4.3.3節では通常の Scheme評価器を修正して ambを実装
する⽅法の詳細を⽰します。

4.3.1 ambと探索
Schemeを拡張して⾮決定性をサポートするようにするために、ambと呼ばれ
る新しい特殊形式を導⼊します。43

(amb ⟨e1⟩ ⟨e2⟩ . . . ⟨en⟩)

この式は、n個の式 ⟨ei ⟩のうちひとつの値を “曖昧に”返します。例えば、次
の式は
(list (amb 1 2 3) (amb 'a 'b))

次の 6つの可能な値を持ちます。
(1 a) (1 b) (2 a) (2 b) (3 a) (3 b)

選択対象がひとつの ambは通常の (ひとつの)値を⽣成します。
選択対象のない amb—式 (amb)—は、受け⼊れ可能な値のない式です。操作的
には、(amb)というものを、評価すると計算を “失敗”させる式と考えることが
できます。計算は異常終了し、値は⽣成されません。この考え⽅を使うと、あ
る特定の述語式 pが真でなければならないことを次のように表現できます。
(define (require p) (if (not p) (amb)))

ambと requireがあれば、上で使った an-element-ofを実装できます。
(define (an-element-of items)

(require (not (null? items)))
(amb (car items) (an-element-of (cdr items))))

an-element-ofはリストが空であれば失敗します。空でなければ、リストの最
初の要素か、リストの残りの要素から選択された要素かのどちらかを曖昧に返
します。
無限の範囲を持つ選択を表現することもできます。次の⼿続きは、ある与

えられた n以上の任意の整数を潜在的に返します。
43⾮決定性プログラミングのための ambという考え⽅は、1961年に John McCarthy
によって初めて書かれたものです (McCarthy 1963参照)。

(define (an-integer-starting-from n)
(amb n (an-integer-starting-from (+ n 1))))

これは、3.5.2節で説明したストリーム⼿続き integers-starting-fromに似て
いますが、ひとつ重要な違いがあります。ストリーム処理は nで始まるすべて
の整数の列を表すオブジェクトを返すのに対し、amb⼿続きはひとつの整数を
返します。44

抽象的には、amb式を評価すると時間の分岐が起こり、それぞれの分岐では
その式の可能な値のひとつによって計算が進むと考えることができます。amb
は ⾮決定性選択点 (nondeterministic choice point)を表現しているという⾔い
⽅をします。動的割り当ての可能な⼗分な数のプロセッサのあるコンピュータ
を持っているとしたら、探索を素直なやり⽅で実装できるところです。amb式
にぶつかるまでは、実⾏は逐次的な機械と同じように進みます。amb式にぶつ
かった時点で、より多くのプロセッサが割り当て・初期化され、その選択の⽰
すすべての並列実⾏が続けられます。それぞれのプロセッサは、失敗によって
終了させられたり、さらに分岐したり、終了したりするまで、その選択が唯⼀
の選択であったかのように計算を続けます。45

⼀⽅、もしひとつのプロセス (または、少しの並⾏プロセス)しか実⾏でき
ない機械を持っている場合は、逐次的な別のやり⽅を考える必要があります。
例えば、評価器を修正して、選択点にぶつかるたびにランダムな分岐を選ぶよ
うにすることも考えられるでしょう。しかし、ランダムに選んで、失敗する値
が多かったとしたらどうでしょう。評価器を何度も何度も動かして、ランダム
な選択をし続けて、失敗しない値を⾒つけるのを願うという⽅法もありますが、
それよりはすべての可能な実⾏パスを 体系的に探索 (systematically search)し
たほうがよいでしょう。この節で開発し、実験を⾏う amb評価器は、体系的探

44実際は、⾮決定的にひとつの選択を返すこととすべての選択を返すこととの違いは、
⾒⽅によるところがあります。値を使うコードの視点からは、⾮決定的な選択はひとつ
の値を返すものです。コードを設計するプログラマの視点からは、⾮決定的な選択とは、
すべての可能な値を潜在的に返し、そこから計算が分岐して、すべての値が別々に調査
されるというものです。

45これは絶望的に⾮効率的な仕組みだと反対する⼈がいるかもしれません。このやり
⽅では、この⽅式で記述された簡単な問題を解くだけでも何百万ものプロセッサが必要
になり、ほとんどの時間ほとんどのプロセッサはアイドル状態になるんじゃないか、と
いう意⾒です。この反対意⾒は歴史的な⽂脈の中で扱われるべきです。以前はメモリは
⾮常に⾼価な商品だと考えられていました。1964年には、1メガバイトの ramが 40万
ドルもしたのです。今では、PCはすべて何メガバイトもの ramを搭載していて、ほと
んどの時間ほとんどのメモリは使われていません。⼤量⽣産される電⼦製品のコストは
思ったよりも安くなりがちなものです。

索を次のように実装します。評価器が ambの適⽤にぶつかると、最初は⼀つ⽬
の選択肢を選びます。この選択は、先でさらに選択肢にぶつかるかもしれませ
ん。評価器は、どの選択点でも常に⼀つ⽬の選択肢を選びます。もしある選択が
失敗に終われば、評価器はオートマジックに 46 直近の選択点にバックトラッ
ク (backtrack)し、次の選択肢を試します。ある選択点で残っている選択肢がな
くなったら、評価器はその前の選択点に戻ってそこから再開します。この処理
は、深さ優先探索 (depth-first search)や 時間的バックトラック (chronological
backtracking)として知られる探索戦略に通じます。47

46automagically: “⾃動的に、しかし何らかの理由で (⼀般的には複雑すぎたり、汚す
ぎたり、時には明⽩すぎたりして)話者が説明する気にならないようなやり⽅で” (Steele
et al. 1983, Raymond 1993)

47⾃動探索戦略をプログラミング⾔語に統合するということには、⻑い波瀾万丈の歴史
があります。探索と⾃動バックトラックによって⾮決定性アルゴリズムをエレガントなや
り⽅でプログラミング⾔語に組み込むことができるのではないかという考えは、Robert
Floyd (1967)が初めて提案したものです。Carl Hewitt (1969)は、⾃動的時間的バックト
ラックを明⽰的にサポートし、組み込みの深さ優先探索戦略を持つ Plannerというプロ
グラミング⾔語を発明しました。Sussman et al. (1971)はこの⾔語のサブセットとなる
MicroPlannerと呼ばれる⾔語を実装し、これは問題解決とロボットの計画の仕事をサポ
ートするのに使われました。また、論理学や定理証明といった分野からも似たような考
えが出てきて、エディンバラとマルセイユで Prologというエレガントな⾔語が⽣み出さ
れることになります (Prologについては4.4節で検討します)。その後、⾃動探索に対し
て多くの不満が出てきた後に、McDermott and Sussman (1972)が Conniverという⾔語
を開発しました。この⾔語は、探索戦略をプログラマのコントロール下に置くための仕
組みを備えていました。しかし、これは扱いにくいということがわかり、Sussman and
Stallman 1975が電⼦回路の記号分析の⽅法を調査している間により扱いやすいアプロ
ーチを発⾒しました。彼らは、事実と事実をつなげる論理的依存性を追跡することによ
る⾮時間的バックトラック計画を開発しました。この技術は、依存主導バックトラック
(dependency-directed backtracking)として知られるようになりました。この⼿法は複雑
でしたが、冗⻑な探索がほとんどないため、それなりに効率のよいプログラムを⽣成し
ました。Doyle (1979)とMcAllester (1978; 1980)は Stallmanと Sussmanの⼿法を⼀般
化・明確化し、探索を定式化する新しいパラダイムを開発しました。これは今では真理
維持 (truth maintenance)と呼ばれています。現代的な問題解決システムはすべて、何ら
かの形で真理維持システムを基層として利⽤しています。真理維持システムや真理維持
を使ったアプリケーションを構築するエレガントな⽅法についての議論は、Forbus and
deKleer 1993を参照してください。Zabih et al. 1987は、ambに基づく Schemeの⾮決定
性拡張について記述しています。これはこの節で説明しているインタプリタに似ていま
すが、時間的バックトラックではなく依存主導バックトラックを使っているため、より
⾼度なものとなっています。Winston 1992には、これら⼆種類のバックトラック両⽅に
ついての紹介があります。

ドライバループ
amb評価器のドライバループには、変わった性質がいくつかあります。このド
ライバループは、式を読み込み、上の prime-sum-pairの例のように、最初の
失敗しなかった実⾏の値を表⽰します。次の成功した実⾏の値を⾒たければ、
バックトラックして⼆番⽬の失敗しなかった実⾏を⽣成するよう試みることを
インタプリタに依頼できます。それには記号 try-againを⼊⼒します。try-
again以外の任意の式が与えられると、インタプリタは前の問題の調べていな
い選択肢を捨て、新しい問題を開始します。以下に対話例を⽰します。
;;; Amb-Eval input:
(prime-sum-pair '(1 3 5 8) '(20 35 110))
;;; Starting a new problem
;;; Amb-Eval value:
(3 20)

;;; Amb-Eval input:
try-again
;;; Amb-Eval value:
(3 110)

;;; Amb-Eval input:
try-again
;;; Amb-Eval value:
(8 35)

;;; Amb-Eval input:
try-again
;;; There are no more values of
(prime-sum-pair (quote (1 3 5 8)) (quote (20 35 110)))

;;; Amb-Eval input:
(prime-sum-pair '(19 27 30) '(11 36 58))
;;; Starting a new problem
;;; Amb-Eval value:
(30 11)

練習問題 4.35: 2つの与えられた限界の間の整数を返す⼿続き an-
integer-betweenを書け。これはピタゴラス数、つまり与えられ
た範囲の間の整数の三つ組 (i, j, k)で i ≤ j かつ i2 + j2 = k2 とな
るものを求める⼿続きを実装するのに使える。
(define (a-pythagorean-triple-between low high)

(let ((i (an-integer-between low high)))
(let ((j (an-integer-between i high)))

(let ((k (an-integer-between j high)))
(require (= (+ (* i i) (* j j)) (* k k)))
(list i j k)))))

練習問題 4.36: 練習問題 3.69では、探索対象の整数の⼤きさに上
限を設けることなく “すべての” ピタゴラス数のストリームを⽣
成する⽅法を検討した。練習問題 4.35の⼿続き内の an-integer-
between を単純に an-integer-starting-from で置き換えるとい
うやり⽅が、任意のピタゴラス数を⽣成する⽅法としてなぜ適切
でないのか説明せよ。これを実際に達成する⼿続きを書け (つま
り、try-againを繰り返し⼊⼒すると原理的にはすべてのピタゴ
ラス数を⽣成することになる⼿続きを書け)。
練習問題 4.37: Ben Bitdiddleは、ピタゴラス数を⽣成する以下の
⼿法が練習問題 4.35のものよりも効率的だと主張している。彼は
正しいだろうか。(ヒント：探索すべき可能性の数を考えよ)

(define (a-pythagorean-triple-between low high)
(let ((i (an-integer-between low high))

(hsq (* high high)))
(let ((j (an-integer-between i high)))

(let ((ksq (+ (* i i) (* j j))))
(require (>= hsq ksq))
(let ((k (sqrt ksq)))

(require (integer? k))
(list i j k))))))

4.3.2 ⾮決定性プログラムの例
4.3.3節では、amb評価器の実装について記述します。しかし、まずはそれをど
う使うかについての例をいくつか挙げます。⾮決定性プログラミングの利点は、
探索の実⾏⽅法の詳細を隠すことによって、プログラムをより⾼い抽象レベル
で表現できるということです。

論理パズル
次のパズル (Dinesman 1968から取ったもの)は、単純な論理パズルという⼤き
なジャンルの中の典型的なものです。

Baker, Cooper, Fletcher, Miller, Smithは 5階しかないアパートの
それぞれ別々の階に住んでいます。Bakerは最上階には住んでい
ません。Cooperは最下階には住んでいません。Fletcherは最上階
にも最下階にも住んでいません。Millerは Cooperよりも⾼い階に
住んでいます。Smithは Fletcherのすぐ上やすぐ下の階には住ん
でいません。Fletcherは Cooperのすぐ上やすぐ下の階には住んで
いません。みんなはそれぞれどの階に住んでいるのでしょうか。

すべての可能性を列挙してから与えられた制約を課すことによって、誰がどの
階に住んでいるかを素直なやり⽅で判断できます。

48

(define (multiple-dwelling)
(let ((baker (amb 1 2 3 4 5)) (cooper (amb 1 2 3 4 5))

(fletcher (amb 1 2 3 4 5)) (miller (amb 1 2 3 4 5))
(smith (amb 1 2 3 4 5)))

(require

48このプログラムでは、リストの要素がそれぞれ別々かどうかを判断するのに次の⼿
続きを使っています。
(define (distinct? items)

(cond ((null? items) true)
((null? (cdr items)) true)
((member (car items) (cdr items)) false)
(else (distinct? (cdr items)))))

memberは memqと同じようなものですが、等価性のテストに eq?ではなく equal?を使
うところが違います。

(distinct? (list baker cooper fletcher miller smith)))
(require (not (= baker 5)))
(require (not (= cooper 1)))
(require (not (= fletcher 5)))
(require (not (= fletcher 1)))
(require (> miller cooper))
(require (not (= (abs (- smith fletcher)) 1)))
(require (not (= (abs (- fletcher cooper)) 1)))
(list (list 'baker baker) (list 'cooper cooper)

(list 'fletcher fletcher) (list 'miller miller)
(list 'smith smith))))

式 (multiple-dwelling)を評価すると、次の結果が⽣成されます。
((baker 3) (cooper 2) (fletcher 4) (miller 5) (smith 1))

この簡単な⼿続きはうまくいくのですが、⾮常に時間がかかります。練習問
題 4.39と 練習問題 4.40では、考えられるいくつかの改善⽅法について検討し
ます。

練習問題 4.38: multiple-dwelling⼿続きを修正し、SmithとFletcher
がすぐ上下の階に住んでいないという制約を取り除け。この変更
したパズルにはいくつの解があるだろうか。
練習問題 4.39: multiple-dwelling⼿続き内の制約の順番は解に影
響するだろうか。答えを⾒つけるのにかかる時間には影響するだ
ろうか。もし順番が関係あると考えるのであれば、与えられた問題
の制約の順番を変えることによって得られるより速いプログラム
を⽰せ。もし関係がないと考えるのであれば、そのことを論じよ。
練習問題 4.40: 複数居住問題で、各⼈を各階に割り当てるやり⽅
は、階の割り当てがそれぞれ別々でなければならないという制約
を設定する前と後で、それぞれ何組あるだろうか。各⼈を各階に
割り当てるすべての可能なやり⽅を⽣成してからバックトラック
にそれらを排除させるのは、⾮常に⾮効率的だ。例えば、ほとんど
の制約は、⼀つか⼆つの⼈と階の変数にしか依存しないので、す
べての⼈に対して階が選択される前にその制約を課すことができ
る。それまでの制約によってすでに排除されていない可能性だけ
を⽣成することによってこの問題を解く、ずっと効率のよい⾮決
定性⼿続きを書き、実演せよ。

練習問題 4.41: 複数居住パズルを解く通常の Schemeプログラム
を書け。
練習問題 4.42: 次の “嘘つき”パズル (Phillips 1934より)を解け。
5⼈の⼥⼦⽣徒が試験のために座っている。彼⼥らは、両親が結果
に過度の関⼼を持っていると考えている。そのため彼⼥らは、試
験の結果について家に⼿紙を送る際に、ひとつ本当のことを書き、
ひとつ嘘のことを書くよう申し合わせた。以下に、彼⼥らの⼿紙
のそれに関連する部分を⽰す。

• Betty: “試験は Kittyが 2位。私は残念ながら 3位。”
• Ethel: “喜んで、私トップ取ったよ。Joanが 2位。”
• Joan: “私は 3位で、かわいそうな Ethelは最下位。”
• Kitty: “私は 2位。Maryは 4位しか取れなかったんだ。”
• Mary: “私は 4位。トップは Bettyだったよ。”

実際には、5⼈の⼥の⼦の成績はどういう順番だったのだろうか。
練習問題 4.43: amb評価器を使って次のパズルを解け。49

Mary Ann Mooreの⽗親と彼の四⼈の友達の Downing⼤佐、Hall
⽒、Barnacle Hood卿、Parker博⼠は、それぞれ⼀台ずつクルーザ
ーを持っている。五⼈にはそれぞれ娘が⼀⼈いて、⾃分のクルー
ザーにはほかの誰かの娘の名前をつけている。Barnacle卿のクル
ーザーはGabrielleで、Moore⽒のは Lorna、Hall⽒のは Rosalind
という名前だ。Downing⼤佐のクルーザーMelissaは Barnacle卿
の娘の名前をつけたものだ。Gabrielleの⽗親は、Parker博⼠の娘
の名前をつけたクルーザーを持っている。Lornaの⽗親は誰か。
効率よく動くプログラムを書くよう努めること (練習問題 4.40参
照)。また、もしMary Annの姓がMooreだと知らされていない場
合はいくつの解が存在するだろうか。
練習問題 4.44: 練習問題 2.42では、“8クイーンパズル”という、ど
の 2つも互いに利き筋に⼊らないように 8つのクイーンをチェス
盤に置くパズルについて記述した。このパズルを解く⾮決定性プ
ログラムを書け。

49これは 1960年代に Litton Industriesから出版された “問題のレクリエーション”と
いう本から取ったものです。これは Kansas State Engineer によるものとされています。

⾃然⾔語の構⽂解析
⾃然⾔語を⼊⼒として受け⼊れるよう設計されたプログラムは、普通はまずそ
の⼊⼒を 構⽂解析 (parse)することを試みます。構⽂解析というのは、⼊⼒を
何らかの⽂法構造に当てはめることです。例えば、“The cat eats.”のような冠
詞・名詞・動詞からなる簡単な⽂を認識しようとしているとします。そのよう
な解析を達成するには、個々の単語の品詞を判定できる必要があります。まず
は、単語を分類する数個のリストを作ってみましょう。50

(define nouns '(noun student professor cat class))
(define verbs '(verb studies lectures eats sleeps))
(define articles '(article the a))

また、⽂法 (grammar)も必要です。⽂法とは、より単純な要素から⽂法要素を
組み⽴てる⽅法を記述した規則の集合です。⾮常に単純な⽂法として、⽂は常
に⼆つの要素—名詞句とそれに続く動詞—からなり、名詞句は定冠詞とそれに
続く名詞からなると規定するようなものが考えられます。この⽂法では、“The
cat eats.”という⽂は次のように構⽂解析されます。
(sentence (noun-phrase (article the) (noun cat))

(verb eats))

それぞれの⽂法規則に対して別々の⼿続きを持つ単純なプログラムによって、
このような解析結果を⽣成できます。⽂を解析するには、⼆つの構成要素を識
別し、それら⼆要素のリストに記号 sentenceのタグをつけたものを返します。
(define (parse-sentence)

(list 'sentence
(parse-noun-phrase)
(parse-word verbs)))

名詞句も同じように、定冠詞とそれに続く名詞を⾒つけることによって解析し
ます。
(define (parse-noun-phrase)

(list 'noun-phrase
(parse-word articles)
(parse-word nouns)))

50ここでは、各リストの最初の要素がリストの残りの単語の品詞を⽰すという仕様に
しています。

⼀番下のレベルでは、構⽂解析は、要求される品詞の単語リストに次の未解析
語が⼊っているかどうかを繰り返しチェックするということになります。これ
を実装するために、未解析の⼊⼒を持つグローバル変数 *unparsed*を保持し
ます。単語をチェックするたびに、*unparsed*が空でないこと、また指定され
たリストの単語から始まることを要求します。要求通りであれば、その単語を
*unparsed*から削除し、単語とその品詞 (リストの先頭にあります)と⼀緒に
返します。51

(define (parse-word word-list)
(require (not (null? *unparsed*)))
(require (memq (car *unparsed*) (cdr word-list)))
(let ((found-word (car *unparsed*)))

(set! *unparsed* (cdr *unparsed*))
(list (car word-list) found-word)))

解析を始めるには、*unparsed*を⼊⼒全体に設定し、⽂の解析を試み、残って
いるものがないことをチェックするだけで⼤丈夫です。
(define *unparsed* '())
(define (parse input)

(set! *unparsed* input)
(let ((sent (parse-sentence)))

(require (null? *unparsed*)) sent))

これで、構⽂解析器を試し、先ほどの簡単なテスト⽂に対してうまく動くか確
認できます。
;;; Amb-Eval input:
(parse '(the cat eats))
;;; Starting a new problem
;;; Amb-Eval value:

(sentence (noun-phrase (article the) (noun cat)) (verb eats))

amb評価器は、構⽂解析上の制約を requireの助けを借りて便利に表現す
るできるという意味で役に⽴っています。しかし、⾃動探索とバックトラック

51parse-wordが未解析の⼊⼒リストを変更するのに set!を使っていることに注意し
てください。これをうまく動作させるには、amb評価器がバックトラックする際に、set!
演算の効果を元に戻す必要があります。

が本当に効果を上げるのは、要素を分解する⽅法に複数の選択肢があるような
より複雑な⽂法について考える場合です。
この⽂法に前置詞を追加してみましょう。

(define prepositions '(prep for to in by with))

そして、前置詞句 (例えば “for the cat”)を前置詞に名詞句が続くものとして定
義します。
(define (parse-prepositional-phrase)

(list 'prep-phrase
(parse-word prepositions)
(parse-noun-phrase)))

⽂は名詞句に動詞句が続くものと定義できます。動詞句は、動詞か、または前
置詞句が後ろにくっついた動詞句のどちらかであるとします。52

(define (parse-sentence)
(list 'sentence (parse-noun-phrase) (parse-verb-phrase)))

(define (parse-verb-phrase)
(define (maybe-extend verb-phrase)

(amb verb-phrase
(maybe-extend
(list 'verb-phrase

verb-phrase
(parse-prepositional-phrase)))))

(maybe-extend (parse-word verbs)))

ついでに、名詞句の定義にも⼿を加えて、“a cat in the class”のようなものを
認めるようにします。ここまで名詞句と呼んできたものは、ここからは単純な
名詞句と呼ぶことにします。そして、今後は名詞句は、単純な名詞句か、前置
詞句が後ろにくっついた名詞句のどちらかということにします。
(define (parse-simple-noun-phrase)

(list 'simple-noun-phrase
(parse-word articles)
(parse-word nouns)))

52この定義が再帰的であることに注意してください。動詞には、任意の数の前置詞句が
続けられることになります。

(define (parse-noun-phrase)
(define (maybe-extend noun-phrase)

(amb noun-phrase
(maybe-extend
(list 'noun-phrase

noun-phrase
(parse-prepositional-phrase)))))

(maybe-extend (parse-simple-noun-phrase)))

この新しい⽂法を使うと、より複雑な⽂を構⽂解析できます。例えば、次の⽂
を解析すると、
(parse '(the student with the cat sleeps in the class))

以下の結果となります。
(sentence
(noun-phrase
(simple-noun-phrase (article the) (noun student))
(prep-phrase
(prep with)
(simple-noun-phrase (article the) (noun cat))))

(verb-phrase
(verb sleeps)
(prep-phrase
(prep in)
(simple-noun-phrase (article the) (noun class)))))

ある⼊⼒に対して⼆つ以上の合法な解析結果がある可能性があるというところ
に注意してください。⽂ “The professor lectures to the student with the cat”
では、教授が猫と⼀緒に講義をしているのかもしれませんし、学⽣が猫を持っ
ているのかもしれません。私たちの⾮決定性プログラムは、両⽅の可能性を⾒
つけます。
(parse '(the professor lectures to the student with the cat))

これを実⾏すると、以下の結果となります。
(sentence
(simple-noun-phrase (article the) (noun professor))

462



(verb-phrase
(verb-phrase
(verb lectures)
(prep-phrase
(prep to)
(simple-noun-phrase (article the) (noun student))))

(prep-phrase
(prep with)
(simple-noun-phrase (article the) (noun cat)))))

評価器に再試⾏を依頼すると、以下の結果となります。
(sentence
(simple-noun-phrase (article the) (noun professor))
(verb-phrase
(verb lectures)
(prep-phrase
(prep to)
(noun-phrase
(simple-noun-phrase (article the) (noun student))
(prep-phrase
(prep with)
(simple-noun-phrase (article the) (noun cat)))))))

練習問題 4.45: 上の⽂法では、次の⽂の構⽂解析結果には 5通り
ある。“The professor lectures to the student in the class with the
cat”。5通りの解析結果を⽰し、それぞれのニュアンスの違いを説
明せよ。
練習問題 4.46: 4.1節と4.2節の評価器は、被演算⼦の評価順を決
めていなかった。amb評価器は被演算⼦を左から右に評価するとい
うことを今後⾒ていく。被演算⼦の評価順がそれと違うものであ
ったら我々の構⽂解析器がうまく動作しないのはなぜか説明せよ。
練習問題 4.47: Louis Reasonerは、動詞句は、動詞か、前置詞句が
後ろに続く動詞句なのだから、⼿続き parse-verb-phraseは次の
ように定義するほうがずっと素直なのではないかと⾔う (名詞句
も同じようにする)。

463



(define (parse-verb-phrase)
(amb (parse-word verbs)

(list 'verb-phrase
(parse-verb-phrase)
(parse-prepositional-phrase))))

これはうまく動くだろうか。amb内の式の順番を⼊れ替えたら、プ
ログラムのふるまいは変わるだろうか。
練習問題 4.48: 上の⽂法を拡張し、より複雑な⽂が扱えるように
せよ。例えば、形容詞や副詞を含むように名詞句や動詞句を拡張
してもよい。または、重⽂を取り扱えるようにしてもよい。53

練習問題 4.49: Alyssa P. Hackerは、⽂を構⽂解析するより、⾯⽩
い⽂を⽣成することに興味がある。彼⼥は、⼿続き parse-wordを
変更して “⼊⼒⽂”を無視するようにして、常に成功して適切な単
語を⽣成するようにするだけで、構⽂解析のために作ったこのプ
ログラムを使って⽂の⽣成ができるのではないかと考えた。Alyssa
の考えを実装し、⽣成された最初の数個の⽂を⽰せ。54

4.3.3 amb評価器の実装
通常の Schemeの式を評価すると、値を返すか、ずっと停⽌しないか、エラー
を発⽣させるかのどちらかになります。⾮決定性 Schemeでは、式を評価する
と、それに加えて⾏き⽌まりの発⾒という結果になることもあります。その場
合、評価は前の選択点にバックトラックしなければならないことになります。
⾮決定性 Schemeの解釈は、このケースが加わるため、複雑になります。

53この種の⽂法は、いくらでも複雑なものにできます。しかし、実際の⾔語理解を考え
ると、このような⽂法はおもちゃにすぎません。実際の⾃然⾔語をコンピュータで理解
するためには、構⽂解析と意味解釈を巧妙に組み合わせる必要があります。⼀⽅、おも
ちゃの構⽂解析器であっても、情報検索システムのようなプログラムの柔軟なコマンド
⾔語をサポートするのには役に⽴ちます。Winston 1992では、コンピュータによる⾃然
⾔語理解のアプローチと、簡単な⽂法のコマンド⾔語への応⽤について論じています。

54Alyssaのアイデアはちゃんと動きます (また、驚くほど単純です)が、⽣成される⽂
はちょっと退屈です—⽣成される⽂は、この⾔語で可能な⽂の中から興味深いものをサ
ンプリングしたものにはなっていないからです。実際のところ、この⽂法はいろいろなと
ころでかなり再帰的で、Alyssaのやり⽅ではそれらの再帰のひとつに “はまって”抜け出
せなくなってしまうのです。これに対処するひとつの⽅法については、練習問題 4.50を
参照してください。

⾮決定性 Schemeのための amb評価器は、4.1.7節の解析評価器を修正して
構築することにします。55 解析評価器と同じように、式の評価は、式の解析によ
って⽣成される実⾏⼿続きを呼ぶことによって達成されます。通常の Scheme
の解釈と⾮決定性 Schemeの解釈の違うところは、完全に実⾏⼿続きの中だけ
に存在します。

実⾏⼿続きと継続
通常の評価器の実⾏⼿続きは、実⾏環境というひとつの引数を取るようになっ
ていました。それに対して、amb評価器の実⾏⼿続きは三つの引数を取るよう
になっています。環境と、継続⼿続き (continuation procedure)と呼ばれる⼆つ
の⼿続きです。式の評価は、これらの⼆つの継続のうちひとつを呼んで終わり
ます。評価の結果が値になったら、成功継続 (success continuation)がその値
で呼ばれます。評価の結果が⾏き⽌まりの発⾒になったら、失敗継続 (failure
continuation)が呼ばれます。継続を適切に構築して呼び出すことが、⾮決定性
評価器がバックトラックを実装する仕組みになっています。
値を受け取り計算を続⾏するのが成功継続の仕事です。成功継続には、値

のほかにまた別の失敗継続が渡されます。これは、値を使った結果が⾏き⽌ま
りになった場合に呼ばれることになります。
⾮決定性処理のほかの分岐を試すのは失敗継続の仕事です。⾮決定性⾔語

の本質は、式が分岐の選択を表現しうるというところにあります。そのような
式を評価すると、どの選択肢が受け⼊れられる結果になるか前もってわからな
くても、⽰された選択肢のひとつに進まなければなりません。これに対処する
ために、評価器はひとつの選択肢を選び、その値を成功継続に渡します。それ
に加えて、評価器は失敗継続を構築し、値と⼀緒に渡します。この失敗継続を
後で呼ぶと、別の選択肢を選ぶことができます。
ユーザープログラムが現在の試⾏を明⽰的に拒絶した場合 (例えば、require

を呼び出すと (amb)という常に失敗する式が実⾏されることになります—4.3.1
節参照)、評価の間に失敗がトリガーされることになります。その時点で⼿元に
ある失敗継続が、直近の選択点にほかの選択肢を選ばせます。その選択点に考
慮すべき選択肢が残っていなければ、それより前の選択点の失敗がトリガーさ
れ . . .という繰り返しになります。失敗継続は、try-again要求に応じて、式の

554.2節の遅延評価器を実装したときは、4.1.1節の通常のメタ循環評価器を修正する
よう選択しました。それに対して、amb評価器は4.1.7節の解析評価器をベースにするこ
とにします。この評価器内の実⾏⼿続きは、バックトラックを実装するのに便利な枠組
みとなっているからです。

ほかの値を⾒つけるためにドライバループから呼び出されることもあります。
それに加えて、選択の結果としての処理の分岐で (変数への代⼊のような)

副作⽤のある演算が起こった場合、処理が⾏き⽌まりを⾒つけたときに、新し
い選択をする前に副作⽤を取り消す必要があるかもしれません。これは、副作
⽤を持つ演算に、副作⽤を取り消し失敗を伝播させる失敗継続を⽣成させるこ
とによって達成できます。
まとめると、失敗継続は以下のように構築されることになります。
• amb式—amb式によって⾏われた現在の選択が⾏き⽌まりに当たったと
きに別の選択を⾏う仕組みを提供します。

• トップレベルドライバ—選択肢を使い果たしたときに失敗を報告する仕
組みを提供します。

• 代⼊—失敗に割り込み、バックトラックの間に代⼊を取り消します。
失敗が起動されるのは、⾏き⽌まりにぶつかった場合だけです。これが起こる
のは、以下のような場合です。

• ユーザープログラムが (amb)を実⾏する。
• ユーザーがトップレベルドライバで try-againを⼊⼒する。

失敗継続は失敗を処理する間にも呼ばれます。
• 代⼊によって作られた失敗継続が副作⽤の取り消しを終えると、割り込
んだ失敗継続を呼び出し、この代⼊に⾄った選択点かトップレベルまで
失敗を伝播させます。

• ある ambの失敗継続が選択肢を使い切ると、その ambに元々渡されてい
た失敗継続を呼び出し、前のチェックポイントかトップレベルまで失敗
を伝播させます。

評価器の構造
amb評価器の⽂法とデータを表現する⼿続きと、基本的な analyze⼿続きは、
4.1.7節の評価器のものとまったく同じものです。違うのは、特殊形式 ambを認
識できるように追加の構⽂⼿続きが必要になるということだけです。56

56評価器は、⾮決定性プログラムで使った let(練習問題 4.22参照)をサポートしてい
ると想定しています。

(define (amb? exp) (tagged-list? exp 'amb))
(define (amb-choices exp) (cdr exp))

analyzeのディスパッチにも、この特殊形式を認識して適切な実⾏⼿続きを⽣
成する節を追加する必要があります。
((amb? exp) (analyze-amb exp))

トップレベル⼿続き ambeval(4.1.7節版の evalに似たもの)は、与えられた式
を解析し、与えられた環境と⼆つの与えられた継続に結果となる実⾏⼿続きを
適⽤します。
(define (ambeval exp env succeed fail)

((analyze exp) env succeed fail))

成功継続は 2引数の⼿続きです。引数のひとつはちょうど得られたばかりの値
で、もうひとつは値が失敗に終わった場合に使うことになるまた別の失敗継続
です。失敗継続は無引数の⼿続きです。つまり、実⾏⼿続きの⼀般的な形は次
のようになります。
(lambda (env succeed fail)

;; succeed is (lambda (value fail) . . .)
;; fail is (lambda () . . .)
. . .)

例えば、以下を実⾏すると、
(ambeval ⟨exp⟩

the-global-environment
(lambda (value fail) value)
(lambda () 'failed))

与えられた式の評価を試み、式の値 (評価が成功した場合) か記号 failed(評
価が失敗した場合) のどちらかを返します。下に⽰す、ドライバループでの
ambevalの呼び出しは、ループを継続し try-again要求をサポートするずっと
複雑な継続⼿続きを使います。

amb評価器の複雑なところは⼤部分、複数の実⾏⼿続きがお互いを呼び出
す間、継続を持ち回らないといけないところから来るものです。以下のコード
を読むときには、それぞれの実⾏⼿続きを4.1.7節の通常の評価器でそれに対
応する⼿続きと⽐較してください。

単純な式
⼀番単純なタイプの式の実⾏⼿続きは、本質的には通常の評価器のものと同じ
になりますが、違うのは継続を管理する必要があるというところです。実⾏⼿
続きは単純に成功し、その式の値と、それとともに⾃分に渡された失敗継続を
返します。
(define (analyze-self-evaluating exp)

(lambda (env succeed fail)
(succeed exp fail)))

(define (analyze-quoted exp)
(let ((qval (text-of-quotation exp)))

(lambda (env succeed fail)
(succeed qval fail))))

(define (analyze-variable exp)
(lambda (env succeed fail)

(succeed (lookup-variable-value exp env) fail)))
(define (analyze-lambda exp)

(let ((vars (lambda-parameters exp))
(bproc (analyze-sequence (lambda-body exp))))

(lambda (env succeed fail)
(succeed (make-procedure vars bproc env) fail))))

変数の検索はいつも ‘成功’ するというところに注意してください。lookup-
variable-valueが変数を⾒つけるのに失敗すると、通常通りエラーを発⽣さ
せます。そのような “失敗”は、プログラムのバグ —未束縛変数への参照—を
⽰しています。これは、いま試している⾮決定性選択肢の代わりに別の選択肢
を試すべきだと⽰しているわけではありません。

条件⽂と列
条件⽂も、通常の評価器の場合と同じように扱われます。analyze-ifによっ
て⽣成される実⾏⼿続きは、述語実⾏⼿続きの pprocに成功継続を渡して起動
します。この成功継続は、述語の値が真かどうかをチェックし、それに応じて
結果部または代替部の実⾏に移ります。pprocの実⾏が失敗した場合は、if式
に対する元の失敗継続が呼ばれます。
(define (analyze-if exp)

(let ((pproc (analyze (if-predicate exp)))
(cproc (analyze (if-consequent exp)))
(aproc (analyze (if-alternative exp))))

(lambda (env succeed fail)
(pproc env

;; pred-valueを得るための
;; 述語の評価に対する成功継続
(lambda (pred-value fail2)

(if (true? pred-value)
(cproc env succeed fail2)
(aproc env succeed fail2)))

;; 述語の評価に対する失敗継続
fail))))

列も、これまでの評価器と同じように扱われます。違うのは、継続を渡すため
に必要となる内部⼿続き sequentiallyの中での仕掛けだけです。その仕掛け
というのは、a, bと順番に実⾏していくときに、bを呼び出す成功継続を渡し
て aを呼び出すというものです。
(define (analyze-sequence exps)

(define (sequentially a b)
(lambda (env succeed fail)

(a env
;; aの呼び出しの成功継続
(lambda (a-value fail2)

(b env succeed fail2))
;; aの呼び出しの失敗継続
fail)))

(define (loop first-proc rest-procs)
(if (null? rest-procs)

first-proc
(loop (sequentially first-proc

(car rest-procs))
(cdr rest-procs))))

(let ((procs (map analyze exps)))
(if (null? procs)

(error "Empty sequence: ANALYZE"))

(loop (car procs) (cdr procs))))

定義と代⼊
定義の場合も、継続の管理にひと⼿間かけなければいけません。これは、実際
に新しい変数を定義する前に、定義の値の式を評価する必要があるからです。
これを達成するために、定義の値の実⾏⼿続き vprocが環境、成功継続、失敗
継続とともに呼び出されます。vprocの実⾏が成功して、定義される変数の値
valが得られたら、その変数が定義され、成功が伝播されます。
(define (analyze-definition exp)

(let ((var (definition-variable exp))
(vproc (analyze (definition-value exp))))

(lambda (env succeed fail)
(vproc env

(lambda (val fail2)
(define-variable! var val env)
(succeed 'ok fail2))

fail))))

代⼊はもっと⾯⽩くなります。ここで初めて、継続を持ち回るだけではなく、
ちゃんと利⽤することになります。代⼊の実⾏⼿続きは、最初のほうは定義の
実⾏⼿続きと似ています。まず、変数に代⼊する新しい値を得ることを試みま
す。この vprocの評価が失敗した場合は、代⼊は失敗します。
しかし、vprocが成功し、代⼊の実⾏に進む場合は、この計算の分岐が後に

なってから失敗する可能性について考えなければなりません。その場合、その
代⼊から抜け出してバックトラックすることが求められます。そのため、バッ
クトラック処理の⼀部として、その代⼊を取り消す準備をしておく必要があり
ます。57

これは、新しい値を変数に代⼊して先に進む前に、古い値を保存する成功継
続 (下のコメント “*1*”の箇所)を vprocに渡すことによって実現します。代
⼊する値とともに渡された失敗継続 (下のコメント “*2*”の箇所)は、失敗を継
続する前に、変数の古い値を復元します。つまり、成功する代⼊は、それに続
く失敗に割り込む失敗継続を提供するということです。fail2を呼んでいたは

57定義の場合、取り消すことについては考えませんでした。内部定義は、⾛査によって
消去されると想定できるためです。(4.1.6節)

ずの失敗はこの⼿続きを代わりに呼び、実際に fail2を呼ぶ前に代⼊を取り消
します。
(define (analyze-assignment exp)

(let ((var (assignment-variable exp))
(vproc (analyze (assignment-value exp))))

(lambda (env succeed fail)
(vproc env

(lambda (val fail2) ; *1*
(let ((old-value

(lookup-variable-value var env)))
(set-variable-value! var val env)
(succeed 'ok

(lambda () ; *2*
(set-variable-value!
var old-value env)

(fail2)))))
fail))))

⼿続きの適⽤
適⽤の実⾏⼿続きには新しい考え⽅はありませんが、違うのは継続を管理する
ことの技術的な複雑さという点です。この複雑さは、analyze-applicationの
中で、被演算⼦を評価する際に成功継続と失敗継続を管理しないといけないと
いうところから来ています。被演算⼦のリストの評価には、通常の評価器のよ
うに単純な mapを使うのでなく、⼿続き get-argsを使います。
(define (analyze-application exp)

(let ((fproc (analyze (operator exp)))
(aprocs (map analyze (operands exp))))

(lambda (env succeed fail)
(fproc env

(lambda (proc fail2)
(get-args aprocs

env
(lambda (args fail3)

(execute-application

proc args succeed fail3))
fail2))

fail))))

get-argsの中では、リスト中のそれぞれの aprocを、再帰的に get-argsを呼
ぶ成功継続とともに呼ぶことによって、aproc実⾏⼿続きリストの cdrダウン
と、結果の argsのリストの consアップを達成していることに注意してくださ
い。これらの再帰的な get-argsの呼び出しはそれぞれ、集積された引数リス
トに新しく得られた引数を consした値を返す成功継続を持っています。
(define (get-args aprocs env succeed fail)

(if (null? aprocs)
(succeed '() fail)
((car aprocs)
env
;; この aprocの成功継続
(lambda (arg fail2)

(get-args
(cdr aprocs)
env
;; get-argsの再帰的呼び出しのための
;; 成功継続
(lambda (args fail3)

(succeed (cons arg args) fail3))
fail2))

fail)))

execute-applicationによって実⾏される実際の⼿続き適⽤は、継続を管理す
る必要があるということ以外は、通常の評価器と同じように達成されます。
(define (execute-application proc args succeed fail)

(cond ((primitive-procedure? proc)
(succeed (apply-primitive-procedure proc args)

fail))
((compound-procedure? proc)
((procedure-body proc)
(extend-environment
(procedure-parameters proc)

args
(procedure-environment proc))

succeed
fail))

(else
(error "Unknown procedure type: EXECUTE-APPLICATION"

proc))))

amb式の評価
特殊形式 ambは、⾮決定性⾔語の鍵となる要素です。ここでは、解釈処理の本
質と、継続を管理する理由について学ぶことになります。ambの実⾏⼿続きは、
amb式のすべての可能な値に対する実⾏⼿続きを巡回する try-nextというル
ープを定義します。それぞれの実⾏⼿続きは、次の実⾏⼿続きを試す失敗継
続とともに呼ばれます。試⾏する選択肢がなくなったら、amb式全体が失敗し
ます。
(define (analyze-amb exp)

(let ((cprocs (map analyze (amb-choices exp))))
(lambda (env succeed fail)

(define (try-next choices)
(if (null? choices)

(fail)
((car choices)
env
succeed
(lambda () (try-next (cdr choices))))))

(try-next cprocs))))

ドライバループ
amb評価器のドライバループは、式の評価の再試⾏をユーザーに許す仕組みの
ために複雑になっています。ドライバは、引数として try-againという⼿続
きを取る internal-loopと呼ばれる⼿続きを使います。これの意図するとこ
ろは、try-againを呼び出すと、⾮決定性評価での次のまだ試していない選択
肢に進まなければいけないということです。internal-loopは、ユーザーがド

ライバループで try-againと⼊⼒した場合には応答として try-againを呼び、
それ以外の場合には ambevalを呼んで新しい評価を開始します。
この ambeval呼び出しに対する失敗継続は、それ以上値がないということ

をユーザーに伝え、ドライバループを再起動するというものです。
ambeval呼び出しに対する成功継続はより難解です。ここでは、得られた

値を表⽰してから、次の選択肢を試⾏できる try-again⼿続きを引数として内
部ループを再起動します。この next-alternative⼿続きは、成功継続に⼆つ
⽬の引数として渡されたものです。通常であれば、この⼆つ⽬の引数は、現在
の評価の分岐が後に失敗した場合に使われる失敗継続として考えます。しかし
この場合は、評価は成功して終わっているので、“失敗”の別の分岐を起動する
ことによって成功する評価を追加で検索できるということになります。
(define input-prompt ";;; Amb-Eval input:")
(define output-prompt ";;; Amb-Eval value:")

(define (driver-loop)
(define (internal-loop try-again)

(prompt-for-input input-prompt)
(let ((input (read)))

(if (eq? input 'try-again)
(try-again)
(begin

(newline) (display ";;; Starting a new problem ")
(ambeval
input
the-global-environment
;; ambeval success
(lambda (val next-alternative)

(announce-output output-prompt)
(user-print val)
(internal-loop next-alternative))

;; ambeval failure
(lambda ()

(announce-output
";;; There are no more values of")

(user-print input)
(driver-loop)))))))

(internal-loop
(lambda ()

(newline) (display ";;; There is no current problem")
(driver-loop))))

internal-loopを最初に呼び出すときには、処理中の問題がないというメ
ッセージを出してドライバループを再起動する try-again⼿続きを使います。
これは、進⾏中の評価がないときにユーザーが try-againを⼊⼒したときに起
こるふるまいです。

練習問題 4.50: 新しい特殊形式 rambを定義せよ。これは ambに似
ているが、次の選択肢を左から右に探索するのではなく、ランダ
ムな順番で探索するというところが違う。これがあると練習問題
4.49の Alyssaの問題を解くのにどう役に⽴つか⽰せ。
練習問題 4.51: permanent-set!という、失敗時に取り消されるこ
とのない新しい種類の代⼊を実装せよ。例えば、以下のように、⼆
つの異なる要素をリストから選択し、成功する選択をするのに要
した試⾏回数を数えるということができる。
(define count 0)
(let ((x (an-element-of '(a b c)))

(y (an-element-of '(a b c))))
(permanent-set! count (+ count 1))
(require (not (eq? x y)))
(list x y count))

;;; Starting a new problem
;;; Amb-Eval value:
(a b 2)
;;; Amb-Eval input:
try-again
;;; Amb-Eval value:
(a c 3)

ここで permanent-set!ではなく set!を使ったら、どのような値
が表⽰されるだろうか。
練習問題 4.52: if-failという、ユーザーが式の失敗を捕捉する
ことを可能にする新しい構造を実装せよ。if-failは⼆つの式を

取る。if-failは最初の式を通常通りに評価し、その評価が成功し
たら通常通りに戻る。しかし、その評価が失敗したら、以下の例の
ように⼆つ⽬の式の値が返される。
;;; Amb-Eval input:
(if-fail (let ((x (an-element-of '(1 3 5))))

(require (even? x))
x)

'all-odd)
;;; Starting a new problem
;;; Amb-Eval value:
all-odd

;;; Amb-Eval input:
(if-fail (let ((x (an-element-of '(1 3 5 8))))

(require (even? x))
x)

'all-odd)
;;; Starting a new problem
;;; Amb-Eval value:
8

練習問題 4.53: 練習問題 4.51で述べた permanent-set!と練習問
題 4.52の if-failがあるとき、以下を評価した結果はどのような
ものになるだろうか。
(let ((pairs '()))

(if-fail
(let ((p (prime-sum-pair '(1 3 5 8)

'(20 35 110))))
(permanent-set! pairs (cons p pairs))
(amb))

pairs))

練習問題 4.54: requireが ambを使う通常の⼿続きとして実装で
きて、ユーザーが⾮決定性プログラムの⼀部として定義できると
いうことに気づいていなければ、requireを特殊形式として実装

する必要があっただろう。そのためには、以下の構⽂⼿続きが必
要になる。
(define (require? exp)

(tagged-list? exp 'require))
(define (require-predicate exp)

(cadr exp))

また、analyze内のディスパッチにも新しい節をひとつ加えなけ
ればならない。
((require? exp) (analyze-require exp))

さらに、require式を扱う⼿続き analyze-requireも必要となる。
以下の analyze-requireの定義を完成させよ。
(define (analyze-require exp)

(let ((pproc (analyze (require-predicate exp))))
(lambda (env succeed fail)

(pproc env
(lambda (pred-value fail2)

(if ⟨??⟩
⟨??⟩
(succeed 'ok fail2)))

fail))))

4.4 論理プログラミング
第 1章では、数学では宣⾔型 (何であるか)の知識を扱うのに対し、計算機科学
では命令型 (どうやるか)の知識を扱うということを強調しました。実際、プロ
グラミング⾔語は、特定の問題を解くためのステップバイステップの⽅法を⽰
す形で知識を表現することをプログラマに求めています。⼀⽅で⾼級⾔語は、
⾔語の実装の⼀部として、相当量の⽅法論的知識が組み込まれているために、
指定した計算がどのように進⾏するかという膨⼤な細かい部分についてプログ
ラマが⼼配しないでもいいようになっています。

Lisp含め、ほとんどのプログラミング⾔語は、数学の関数の値を計算する
ことを中⼼に構成されています。式指向の⾔語 (Lisp、Fortran、Algolなど)は、

関数の値を記述する式が、その値を求める⼿段としても解釈できるという “⾔
葉遊び”を利⽤しています。このため、多くのプログラミング⾔語は、単⽅向
計算 (明確に定義された⼊⼒と出⼒を持つ演算)の⽅向に強く偏っています。し
かし、この偏りを和らげる、根本的に異なるプログラミング⾔語も存在します。
3.3.5節では、そのような⾔語の例をひとつ⾒ました。この⾔語では、計算の
オブジェクトは算術的制約でした。制約システムでは、演算の向きと順番はあ
まりはっきり規定されていません。そのため、計算を実⾏する際には、システ
ムは普通の算術演算の場合よりも、より詳細な “どうやるか”の知識を提供す
る必要がありました。しかしこれは、命令的知識を提供するという責任からユ
ーザーが完全に⾃由になったということではありません。同じ制約の集合を実
装する制約ネットワークは数多くあり、ユーザーは数学的に等価なネットワー
クの集合から特定の計算を規定するのに適したネットワークを選ぶ必要があり
ます。

4.3節の⾮決定性プログラム評価器も、プログラミングとは⼀⽅向性関数の
計算のためのアルゴリズムを構築することであるという視点からは離れていま
す。⾮決定性⾔語では、式は⼆つ以上の値を持つことがありえ、その結果とし
て計算は⼀価関数を扱うというより関係を扱うものとなります。論理プログラ
ミングでは、プログラミングを関係性という視点から⾒ることと、ユニフィケ
ーション (unification)と呼ばれる強⼒な記号パターンマッチングとを組み合わ
せて、この考えを拡張しています。58

58論理プログラミングは、⾃動定理証明の研究の⻑い歴史から派⽣したものです。初
期の定理証明プログラムにできることは⾮常に限られていました。これは、可能な証明
空間を網羅的に探索していたことによるものです。そのような探索が実⽤的になったの
は、1960年代早期に ユニフィケーションアルゴリズム (unification algorithm)と 導出
原理 (resolution principle)(Robinson 1965)の発⾒といった技術⾰新があったことによ
ります。例えば、導出はGreen and Raphael (1968)(Green 1969も参照)で推論的質問応
答システムの基盤として使われています。この時期のほとんどの間、研究者は証明が存
在するならその証明を⾒つけられることが保証されたアルゴリズムに焦点を当てていま
した。そのようなアルゴリズムは、制御して証明に向かわせることが難しいものでした。
Hewitt (1969)は、プログラミング⾔語の制御構造と論理操作システムを融合させる可能
性に気づき、それが 4.3.1節(Footnote 4.47)で述べたような⾃動探索の研究につながり
ました。これと同じ時期に、マルセイユの Colmerauerは⾃然⾔語を扱うルールベース
システム (Colmerauer et al. 1973)を開発していました。彼はこれらのルールを表現す
るために Prologというプログラミング⾔語を発明しました。エディンバラのKowalski
(1973; 1979)は、Prologプログラムの実⾏は (線形ホーン節導出という証明テクニックを
使って)定理証明として解釈できるということに気づきました。これらの⼆つの流れが合
流して、論理プログラミング運動が起こることになります。このため、論理プログラミ
ングの発展の功労者について考えると、フランスにはマルセイユ⼤学での Prologの誕⽣

このアプローチは、うまくいけば、プログラムを書くとても強⼒な⽅法に
なりえます。この⼒の⼀部は、単⼀の “何であるか”という事実が、いくつも
の “どうやるか”という構成要素を含むような複数の問題を解くのに使えるこ
ともあるということから来ています。例として、⼆つのリストを引数として取
り、それらの要素を結合してひとつのリストを作る append演算について考え
てみましょう。Lispのような⼿続き型⾔語では、2.2.1節でやったように、基本
的なリストコンストラクタの consを使って appendを定義できます。

(define (append x y)
(if (null? x) y (cons (car x) (append (cdr x) y))))

この⼿続きは、次の⼆つの規則によって Lispに翻訳したものと捉えることが
できます。⼀つ⽬の規則は⼀つ⽬のリストが空である場合を扱い、⼆つ⽬の規
則は空でないリスト、つまり⼆つの部分の consである場合を扱います。

• 任意のリスト yについて、空リストと yを appendすると yになる。
• 任意の u, v, y, zについて、vと yを appendすると zになるならば、(cons

u v)と yを appendすると (cons u z)になる。59

append⼿続きを使うと、次のような質問に答えることができます。
(a b)と (c d)の appendを求めよ。

しかし、この⼆つの規則は、そのままで次のような質問に答えるにも⼗分なも
のです。上の append⼿続きはこういった質問には答えられません。

(a b)と appendすると (a b c d)になるようなリスト yを求めよ。
appendすると (a b c d)になるようなすべての xと yを求めよ。

論理プログラミング⾔語では、プログラマは上に挙げた appendについての⼆
つの規則を述べることによって append“⼿続き”を書きます。“どうやって”と
いう知識はインタプリタによって⾃動的に提供され、この⼆つの規則だけを使
という得点がある⼀⽅で、イギリスはエディンバラ⼤学での成果というところをアピー
ルできます。mitの⼈に⾔わせると、論理プログラミングは、Hewittの傑出した、しか
し難解な博⼠論⽂をこれらのグループが解明しようとする中で発展したそうです。論理
プログラミングの歴史については、Robinson 1983を参照してください。

59規則と⼿続きの対応関係を⾒るには、⼿続きの x(xが空でない場合)を規則の (cons
u v)に対応させます。すると、規則の zは (cdr x)と yを appendしたものに対応しま
す。

って appendについての三つのタイプの質問すべてに答えられるようにします。
60

現代の論理プログラミング⾔語 (ここで実装しているものを含め)には、⼀
般的な “どうやって”の⼿法が間違った無限ループなどの望ましくないふるま
いを引き起こすという点で、本質的な⽋点を抱えています。論理プログラミン
グは計算機科学の活発な研究領域となっています。61

この章の初めのほうでは、インタプリタの実装技術について掘り下げ、Lisp
⾵⾔語のインタプリタにとって (というよりも、どんな従来型のインタプリタ
にとっても)本質的なものとなる要素について記述しました。ここからは、あ
る論理プログラミング⾔語のインタプリタについて考えるために、これらの考
えを応⽤していきます。この⾔語は、⾔語で表現したクエリ (query)、つまり
質問を作ることによってデータベースから情報を検索するのにとても便利なの
で、クエリ⾔語 (query language)と呼ぶことにします。クエリ⾔語は Lispとは
まったく違うものなのですが、この⾔語の記述には、これまでずっと使ってき
たものと同じ⼀般的な枠組みがうまく使えるということをこれから⾒ていきま
す。その枠組みというのは、基本要素の集合と、単純な要素を組み合わせてよ
り複雑な要素を作る組み合わせ⼿段と、複雑な要素を単純な概念的単位と捉え
ることを可能にするような抽象化⼿段です。論理プログラミング⾔語のインタ
プリタは、Lispのような⾔語のインタプリタよりもかなり複雑です。それでも、
私たちのクエリ⾔語インタプリタは4.1節のインタプリタにあったものと同じ

60これはどうやって答えを計算するかという問題全体からユーザーを解放してくれる
ものではないことは確かです。appendの関係を定式化する数学的に等価な規則は数多く
ありますが、どちら向きの計算でもできるような有効な⼿段として使えるものはその⼀
部だけです。それに加えて、“何であるか” という情報が “どうやって”答えを計算するか
ということについて何の⼿がかりにもならないということもあります。例えば、y2 = x
となる yを求めるという問題について考えてみてください。

61論理プログラミングに対する関⼼がピークを迎えたのは 80 年代前半のことです。
⽇本政府が論理プログラミング⾔語の実⾏に最適化された超⾼速コンピュータの作成
を⽬指した野⼼的なプロジェクトを始めたのはこのころのことです。このようなコン
ピュータの速度は、通常の FLOPS (FLoating-point Operations Per Second) でなく、
LIPS(Logical Inferences Per Second)で測られるようになっていました。このプロジェ
クトは、ハードウェアとソフトウェアを計画通りに開発するという点では成功したので
すが、国際的なコンピュータ業界は別の⽅向に進んでいました。⽇本のこのプロジェク
トについての概括的評価についてはFeigenbaum and Shrobe 1993を参照してください。
論理プログラミングのコミュニティも、単純なパターンマッチング以外の技術に基づく
関係プログラミングを考慮に⼊れる⽅向に動きました。そのような技術としては、3.3.5
節の制約伝播システムで⽰したような数値制約を取り扱う能⼒のようなものがあります。

要素を多く含んでいます。具体的には、式を型によって分類する “eval”部があ
り、⾔語の抽象化メカニズム (Lispの場合は⼿続きで、論理プログラミングの
場合は規則 (rule))を実装する “apply”部があります。また、中⼼的な役割を果
たすのは、記号とそれに関係づけられた値との対応を決めるフレームデータ構
造による実装です。このクエリ⾔語の実装のもうひとつの⾯⽩い側⾯は、第 3
章で導⼊したストリームが⼤活躍するということです。

4.4.1 推論的情報検索
論理プログラミングは、情報検索のためのデータベースのインターフェイスを
提供するのに優れています。この章で実装するクエリ⾔語は、そのように使う
ことを意図して設計されています。
クエリシステムが何をしているかを説明するために、このシステムを

Microshaftというボストン地区の活発なハイテク企業で⼈事記録を管理するの
に使うという例を⽰すことにします。この⾔語は⼈事情報に対するパターン主
導のアクセスを提供し、また論理的推論を⾏うための⼀般的な規則も利⽤でき
ます。

サンプルデータベース
Microshaftの⼈事情報データベースは、従業員に関する表明 (assertion)を持っ
ています。フルタイムで働くコンピュータウィザード (達⼈)の Ben Bitdiddle
に関する情報は次のようになっています。
(address (Bitdiddle Ben) (Slumerville (Ridge Road) 10))
(job (Bitdiddle Ben) (computer wizard))
(salary (Bitdiddle Ben) 60000)

それぞれの表明はリスト (この場合は三つ組)で、リストの要素はそれ⾃体がリ
ストであることもあります。
フルタイムのウィザードとして、Benは会社のコンピュータ部⾨の責任者

で、プログラマ⼆⼈と技術者⼀⼈を監督しています。彼らの情報は以下の通り
です。
(address (Hacker Alyssa P) (Cambridge (Mass Ave) 78))
(job (Hacker Alyssa P) (computer programmer))
(salary (Hacker Alyssa P) 40000)
(supervisor (Hacker Alyssa P) (Bitdiddle Ben))

(address (Fect Cy D) (Cambridge (Ames Street) 3))
(job (Fect Cy D) (computer programmer))
(salary (Fect Cy D) 35000)
(supervisor (Fect Cy D) (Bitdiddle Ben))

(address (Tweakit Lem E) (Boston (Bay State Road) 22))
(job (Tweakit Lem E) (computer technician))
(salary (Tweakit Lem E) 25000)
(supervisor (Tweakit Lem E) (Bitdiddle Ben))

Alyssaが監督する研修⽣プログラマもいます。
(address (Reasoner Louis) (Slumerville (Pine Tree Road) 80))
(job (Reasoner Louis) (computer programmer trainee))
(salary (Reasoner Louis) 30000)
(supervisor (Reasoner Louis) (Hacker Alyssa P))

これらの⼈々は、職務 (job)記述の第⼀項の単語 computerが⽰すように、みな
コンピュータ部⾨に属しています。

Benは上級社員で、監督者は会社の重役です。
(supervisor (Bitdiddle Ben) (Warbucks Oliver))
(address (Warbucks Oliver) (Swellesley (Top Heap Road)))
(job (Warbucks Oliver) (administration big wheel))
(salary (Warbucks Oliver) 150000)

Benの監督するコンピュータ部⾨のほかに、会社には会計主任とアシスタント
からなる経理部⾨もあります。
(address (Scrooge Eben) (Weston (Shady Lane) 10))
(job (Scrooge Eben) (accounting chief accountant))
(salary (Scrooge Eben) 75000)
(supervisor (Scrooge Eben) (Warbucks Oliver))

(address (Cratchet Robert) (Allston (N Harvard Street) 16))
(job (Cratchet Robert) (accounting scrivener))
(salary (Cratchet Robert) 18000)
(supervisor (Cratchet Robert) (Scrooge Eben))

また、重役の秘書もいます。
(address (Aull DeWitt) (Slumerville (Onion Square) 5))
(job (Aull DeWitt) (administration secretary))
(salary (Aull DeWitt) 25000)
(supervisor (Aull DeWitt) (Warbucks Oliver))

このデータベースは、ほかの職務を持つ⼈に⾏える職務についての表明も⼊っ
ています。例えば、コンピュータウィザードはコンピュータプログラマとコン
ピュータ技術者のどちらの職務も⾏うことができます。
(can-do-job (computer wizard) (computer programmer))
(can-do-job (computer wizard) (computer technician))

コンピュータプログラマには研修⽣の代わりができます。
(can-do-job (computer programmer)

(computer programmer trainee))

また、次のこともよく知られています。訳注 6

(can-do-job (administration secretary)
(administration big wheel))

単純クエリ
このクエリ⾔語は、ユーザーがシステムのプロンプトにクエリを書くことによ
って、データベースから情報を検索できるようになっています。例えば、コン
ピュータプログラマをすべて検索するには、次のようにします。
;;; Query input:
(job ?x (computer programmer))

システムは、次のように⼆つの項⽬を返します。
;;; Query results:
(job (Hacker Alyssa P) (computer programmer))
(job (Fect Cy D) (computer programmer))

訳注 6秘書には重役の仕事ができる、つまり重役は実際のところたいした仕事をしていな
いという⽪⾁です。

⼊⼒クエリは、ある特定のパターン (pattern)にマッチするデータベースのエ
ントリを探しているということを規定しています。この例でパターンが規定し
ているのは、三つの項⽬からなるエントリで、その⼀つ⽬の項⽬は⽂字通りの
記号の job、⼆つ⽬は任意のもので、三つ⽬は (computer programmer)という
⽂字通りのリストというものです。マッチングリストの⼆つ⽬の項⽬となるこ
の “任意のもの”は、パターン変数 (pattern variable)の ?xによって規定されて
います。パターン変数の⼀般的な形式は、変数の名前となる記号の前に ?をつ
けたものです。?を単独でパターンに⼊れて “任意のもの”を表すのではなく、
パターン変数に名前を規定したほうが便利な理由についてはこの先で説明しま
す。システムは、単純クエリに対して、規定されたパターンにマッチするデー
タベースの全エントリを表⽰するという応答を返します。
パターンは複数の変数を持つことができます。例えば、以下のクエリは

(address ?x ?y)

従業員すべての住所を列挙します。
変数を持たないパターンを作ることもできます。その場合、クエリは単純

にそのパターンがデータベース内にエントリとして存在しているかどうかを判
断します。あるなら、マッチがひとつあることになります。なければマッチは
ありません。
ひとつのクエリで同じパターン変数を複数回使うこともできます。これは、

同じ “任意のもの”がそれぞれの場所に現れる必要があるということを規定し
ています。変数が名前を持つのはこのためです。例えば、
(supervisor ?x ?x)

このクエリは、⾃分⾃⾝が監督者となっている⼈々をすべて探します (しかし、
ここでのサンプルデータベースにはそのような表明はありません)。
以下のクエリは、

(job ?x (computer ?type))

三つ⽬の項⽬が⼆要素のリストで、その⼀つ⽬の要素が computerであるすべ
ての職務エントリにマッチします。
(job (Bitdiddle Ben) (computer wizard))
(job (Hacker Alyssa P) (computer programmer))
(job (Fect Cy D) (computer programmer))
(job (Tweakit Lem E) (computer technician))

このパターンは、次のものにはマッチ “しません”。

(job (Reasoner Louis) (computer programmer trainee))

これは、このエントリの三つ⽬の項⽬が三要素のリストであるのに対し、パタ
ーンの三つ⽬の項⽬は⼆要素であるべきだと規定しているためです。三つ⽬の
項⽬が computerで始まる任意のリストとなるようにパターンを変更するには、
以下のように規定します。62

(job ?x (computer . ?type))

例えば、
(computer . ?type)

このクエリは次のデータにマッチします。
(computer programmer trainee)

このとき、?typeはリスト (programmer trainee)になります。このクエリは、
次のデータにもマッチします。
(computer programmer)

このとき、?typeはリスト (programmer)になります。また、次のデータにも
マッチします。
(computer)

このとき、?typeは空リスト ()になります。
クエリ⾔語の簡単な問い合わせ処理は次のように記述できます。
• システムは、クエリパターン内の変数に対して、パターンを満たす

(satisfy) すべての代⼊を探します。これはつまり、パターン変数が値に
よって具体化される (instantiated with)と結果がデータベースに存在す
るようなすべての値の集合を探すということです。

• システムは、パターンを満たす変数代⼊によるクエリパターンの具体化
をすべて列挙することによってクエリに応答します。

パターンに変数がなければ、クエリはそのパターンがデータベースにあるかど
うかを判断するというだけのことになります。データベースにある場合、空の
代⼊、つまり変数に値を代⼊しない代⼊が、そのデータベースに対してパター
ンを満たすことになります。

62ここでは、練習問題 2.20で紹介したドット末尾記法を使っています。

練習問題 4.55: 以下の情報をデータベースから取り出す単純クエ
リを⽰せ。

1. Ben Bitdiddleが監督者となっている⼈すべて
2. 経理部⾨に属するすべての⼈の名前と職務
3. Slumervilleに住むすべての⼈の名前と住所

複合クエリ
単純クエリはクエリ⾔語の基本演算となっています。複合演算を作るために、
クエリ⾔語は組み合わせ⼿段を提供しています。クエリ⾔語を論理プログラミ
ング⾔語といえるものにしている要素のひとつは、組み合わせ⼿段が論理式を
作るのに使う組み合わせ⼿段とそっくりであるという点です。組み合わせ⼿段
は and, or, notの三つです (ここでの and, or, notは Lispの基本演算ではなく、
クエリ⾔語の組み込み演算です)。
andを次のように使うことによって、コンピュータプログラマの全員の住所を
検索できます。
(and (job ?person (computer programmer))

(address ?person ?where))

これの結果は、次のような出⼒となります。
(and (job (Hacker Alyssa P) (computer programmer))

(address (Hacker Alyssa P) (Cambridge (Mass Ave) 78)))
(and (job (Fect Cy D) (computer programmer))

(address (Fect Cy D) (Cambridge (Ames Street) 3)))

⼀般的に、
(and ⟨query1⟩ ⟨query2⟩ . . . ⟨queryn⟩)

この式が満たされるのは、パターン変数に対するすべての値の集合が、同時に
⟨query1⟩ . . . ⟨queryn⟩を満たすときです。
単純クエリの場合と同じように、システムが複合クエリを処理する際には、パ
ターン変数に対する代⼊でそのクエリを満たすものをすべて探し、それらの値
によるクエリの具体化をすべて表⽰します。
複合クエリを構築するもうひとつの⼿段は orによるものです。例えば、

(or (supervisor ?x (Bitdiddle Ben))
(supervisor ?x (Hacker Alyssa P)))

この式は、Ben Bitdiddleまたは Alyssa P. Hackerに監督される従業員をすべ
て検索します。
(or (supervisor (Hacker Alyssa P) (Bitdiddle Ben))

(supervisor (Hacker Alyssa P) (Hacker Alyssa P)))
(or (supervisor (Fect Cy D) (Bitdiddle Ben))

(supervisor (Fect Cy D) (Hacker Alyssa P)))
(or (supervisor (Tweakit Lem E) (Bitdiddle Ben))

(supervisor (Tweakit Lem E) (Hacker Alyssa P)))
(or (supervisor (Reasoner Louis) (Bitdiddle Ben))

(supervisor (Reasoner Louis) (Hacker Alyssa P)))

⼀般的に、
(or ⟨query1⟩ ⟨query2⟩ . . . ⟨queryn⟩)

この式が満たされるのは、パターン変数に対するすべての値の集合が、⟨query1⟩
. . . ⟨queryn⟩ のうち少なくともひとつを満たすときです。
複合クエリは、notによって作ることもできます。例えば、
(and (supervisor ?x (Bitdiddle Ben))

(not (job ?x (computer programmer))))

この式は、Ben Bitdiddleに監督され、コンピュータプログラマではないすべて
の⼈を検索します。⼀般的に、
(not ⟨query1⟩)

この式が満たされるのは、パターン変数に対するすべての代⼊が ⟨query1⟩ を
満たさない場合に満たされます。63

最後の組み合わせ形式は lisp-valueと呼ばれるものです。lisp-valueが
パターンの⼀つ⽬の要素であれば、それは次の要素が Lispの述語であり、残
りの (具体化された) 要素を引数として適⽤するということを規定しています。
⼀般的に、

63実際は、notに対するこの記述が有効なのは簡単な場合だけです。実際の notのふる
まいはより複雑です。notの独特なところについては、4.4.2節と4.4.3節で詳しく⾒てい
きます。

(lisp-value ⟨predicate⟩ ⟨arg1⟩ . . . ⟨argn⟩)

この式を満たすのは、⟨predicate⟩が具体化された ⟨arg1⟩ . . . ⟨argn⟩に適⽤さ
れたときの値が真になるようなパターン変数への代⼊です。例えば、給料が
30,000ドルより多い⼈をすべて検索する場合は、次のように書くことができま
す。64

(and (salary ?person ?amount) (lisp-value > ?amount 30000))

練習問題 4.56: 以下の情報を取得する複合クエリを考えよ。
a. Ben Bitdiddleに監督されるすべての⼈と、その住所
b. Ben Bitdiddleよりも給料が少ないすべての⼈と、その給料
と、Ben Bitdiddleの給料

c. コンピュータ部⾨以外の⼈に監督されているすべての⼈と、
その上司の名前と職務

規則
単純クエリと複合クエリのほかにも、クエリ⾔語はクエリを抽象化する⼿段を
提供しています。その⼿段とは規則 (rule)と呼ばれるものです。以下の規則は、
(rule (lives-near ?person-1 ?person-2)

(and (address ?person-1 (?town . ?rest-1))
(address ?person-2 (?town . ?rest-2))
(not (same ?person-1 ?person-2))))

⼆⼈の⼈が同じ町に住んでいるなら、互いに近くに住んでいる (lives-near)
とする、ということを規定しています。最後の not節は、この規則によってす
べての⼈がその⼈⾃⾝の近くに住んでいるということになるのを防いでいま
す。same関係はとても単純な規則によって定義されています。65

64lisp-valueは、クエリ⾔語によって提供されていない演算を⾏う場合に限って使う
べきです。具体的には、等価性のテストや、⾮等価性のテストに使ってはいけません (等
価性については、クエリ⾔語のマッチがまさにそのために設計されたものだからです。⾮
等価性については、下で⽰す sameによって⾏うことができます)。

65⼆つのものを同じものであるようにするには sameは必要ないということに注意して
ください。その場合は、同じパターン変数をそれぞれに使うだけで⼗分です— つまり、
そもそも⼆つのものを持たずに、ひとつのものしか持たなければいいということです。例

(rule (same ?x ?x))

以下の規則は、ある⼈が誰かを監督する⼈を監督している場合、その⼈を組織
内での “wheel”(重要⼈物)であると宣⾔するものです。
(rule (wheel ?person)

(and (supervisor ?middle-manager ?person)
(supervisor ?x ?middle-manager)))

規則の⼀般形式は次のようになります。
(rule ⟨conclusion⟩ ⟨body⟩)

ここで、⟨conclusion⟩はパターンで、⟨body⟩は任意のクエリです。66 規則は、
⼤きな (無限であることすらある)表明の集合、つまり規則の本体を満たす変数
代⼊によって規則の結論を具体化したものすべてを表現するものとして考える
ことができます。単純クエリ (パターン)について記述したときには、具体化さ
れたパターンがデータベースに存在する場合に変数への代⼊がパターンを満た
すと書きました。しかし、パターンは表明としてデータベースに明⽰的に存在
する必要はありません。それは、規則が含意する暗黙的な表明であることもあ
ります。例えば、以下のクエリは、
(lives-near ?x (Bitdiddle Ben))

次の結果となります。
(lives-near (Reasoner Louis) (Bitdiddle Ben))
(lives-near (Aull DeWitt) (Bitdiddle Ben))

Ben Bitdiddleの近くに住むコンピュータプログラマをすべて検索するには、次
のように質問できます。
(and (job ?x (computer programmer))

(lives-near ?x (Bitdiddle Ben)))

えば、lives-near規則の?townや、以下の wheel規則の ?middle-managerがそうです。
sameが役に⽴つのは、lives-near規則の ?person-1と?person-2のように、⼆つのも
のが違うものであることを強制する場合です。同じパターン変数をクエリの⼆箇所で使
うと、両⽅の場所で同じ値が現れることを強制するのですが、別々のパターン変数を使
っても、別々の値が現れることを強制するわけではありません (別々のパターン変数に
代⼊される値は、同じであることも別々であることもあります)。

66sameのような本体のない規則も認めることにします。そのような規則は、規則の結
論は変数の値がどのようなものであっても満たされているものとして解釈します。

複合⼿続きの場合と同じように、規則はほかの規則の⼀部として (上記の
lives-near規則で⾒たように)使うことができます。さらに、再帰的な定義も
可能です。例えば、次の規則は、
(rule (outranked-by ?staff-person ?boss)

(or (supervisor ?staff-person ?boss)
(and (supervisor ?staff-person ?middle-manager)

(outranked-by ?middle-manager ?boss))))

ある職員 (staff-person)について、ボス (boss)がその⼈の上司であるか、(再
帰的に)その職員の上司よりそのボスが地位が上 (outranked)であれば、その
ボスはその職員より地位が上であるというものです。

練習問題 4.57: ⼈ 1が⼈ 2と同じ職務を持っているか、⼈ 1の職
務を持つ⼈には⼈ 2の職務もできるという場合で、⼈ 1と⼈ 2が
同⼀⼈物でない場合、⼈ 1は⼈ 2を代替 (replace)できるとする
規則を定義せよ。この規則を使って、以下の検索を⾏うクエリを
⽰せ。

a. Cy D. Fectを代替できるすべての⼈
b. ⾃分より給料の⾼い誰かを代替できるすべての⼈と、その⼆
つの給料

練習問題 4.58: ある⼈が、⾃分の勤める部署に勤める監督者がい
ない場合、その⼈を “big shot”(重要⼈物)であるとする規則を定
義せよ。
練習問題 4.59: Ben Bitdiddleはある会議を何度も⽋席してしまっ
た。この会議を忘れる癖を何とかしないと⾸になると思って、Ben
は対策を取ることを決めた。彼は、以下のような表明として会社の
週次ミーティングを Microshaftのデータベースにすべて追加した。
(meeting accounting (Monday 9am))
(meeting administration (Monday 10am))
(meeting computer (Wednesday 3pm))
(meeting administration (Friday 1pm))

それぞれの表明は部⾨の全体会議のものだ。それに加えて、Benは
全部⾨にわたる全社会議のエントリも追加した。この会議には会
社の全従業員が参加する。

490



(meeting whole-company (Wednesday 4pm))

a. ⾦曜の朝、Benはその⽇にあるすべての会議についてデータ
ベースに問い合わせようと思った。彼はどのようなクエリを
使うべきだろうか。

b. Alyssa P. Hackerは、このやり⽅はあまりよくないと思った。
彼⼥は、⾃分の名前を指定して⾃分の会議について聞くこと
ができたほうがずっと便利だろうと考えた。そこで、彼⼥は
規則を設計することにした。その規則は、ある⼈の会議は全
社 (whole-company)会議とその⼈の部⾨会議をすべて含むと
いうものだ。Alyssaの規則に本体を補え。
(rule (meeting-time ?person ?day-and-time)

⟨rule-body⟩)

c. Alyssaは⽔曜の朝に職場に着き、その⽇にどんな会議がある
か考えた。上記の規則を定義してあるとして、この検索を⾏
うには、彼⼥はどのようなクエリを作るべきだろうか。

練習問題 4.60: 以下のクエリによって、
(lives-near ?person (Hacker Alyssa P))

⾃分の近所に住んでいて会社まで⾞の相乗りができる⼈を Alyssa
P. Hackerは探すことができている。⼀⽅、彼⼥が次のクエリを使
って互いに近所に住んでいる⼈のペアをすべて検索しようとした
とき、
(lives-near ?person-1 ?person-2)

互いに近所に住んでいる⼈のペアがそれぞれ⼆回ずつ表⽰されて
いることに気がついた。例えば、次のようなものだ。
(lives-near (Hacker Alyssa P) (Fect Cy D))
(lives-near (Fect Cy D) (Hacker Alyssa P))

これはなぜ起こるのだろうか。それぞれのペアが⼀回ずつしか現
れないようにして、互いに近所に住んでいる⼈のリストを検索す
る⽅法はあるだろうか。説明せよ。

プログラムとしての論理
規則は論理的含意の⼀種と⾒なすことができます。もしパターン変数に対する
値の代⼊が本体を満たすのであれば、その代⼊は結論を満たします。したがっ
て、クエリ⾔語は規則に基づき 論理的推論 (logical deduction)を⾏う能⼒があ
るものと⾒なすことができます。例として、4.4節の初めに記述した append演
算について考えてみましょう。すでに述べたように、appendは次の⼆つの規則
によって特徴づけられます。

• 任意のリスト yについて、空リストと yを appendすると yになる。
• 任意の u, v, y, zについて、vと yを appendすると zになるならば、(cons

u v)と yを appendすると (cons u z)になる。
これを私たちのクエリ⾔語で表現するために、次の関係に対して⼆つの規則を
定義します。
(append-to-form x y z)

この関係は、“xと yを appendすると zになる”ということだと解釈できます。
(rule (append-to-form () ?y ?y))
(rule (append-to-form (?u . ?v) ?y (?u . ?z))

(append-to-form ?v ?y ?z))

最初の規則には本体がありません。これは、?yがどのような値であっても結論
が成り⽴つということです。⼆つ⽬の規則で、リストの carと cdrに名前をつ
けるためにドット末尾記法を使っていることに注意してください。
これら⼆つの規則があれば、⼆つのリストの appendを計算するクエリを定

式化できます。
;;; Query input:
(append-to-form (a b) (c d) ?z)
;;; Query results:
(append-to-form (a b) (c d) (a b c d))

それよりもすごいのは、“(a b)に appendすると (a b c d)になるリストはど
のようなリストか”と質問するのにも同じ規則が使えるというところです。こ
れは、以下のように実⾏できます。
;;; Query input:
(append-to-form (a b) ?y (a b c d))

;;; Query results:
(append-to-form (a b) (c d) (a b c d))

appendすると (a b c d)になるすべてのリストのペアについて質問すること
も可能です。
;;; Query input:
(append-to-form ?x ?y (a b c d))
;;; Query results:
(append-to-form () (a b c d) (a b c d))
(append-to-form (a) (b c d) (a b c d))
(append-to-form (a b) (c d) (a b c d))
(append-to-form (a b c) (d) (a b c d))
(append-to-form (a b c d) () (a b c d))

規則を使って上記のクエリの答えを推論しているという点で、このクエリシス
テムはかなりの知性を⽰しているかのように⾒えるかもしれません。実際のと
ころは、次の節で⾒ていくように、このシステムは規則を解きほぐすにあたっ
て明確に定義されたアルゴリズムに従っています。残念ながら、このシステム
は appendの場合には⾒事なぐらいうまくいくのですが、より複雑な場合には、
4.4.3節で⾒ていくように、⼀般的な⽅法が破綻することもあります。

練習問題 4.61: 次の⼆つの規則は、リストの隣り合う要素を求め
る next-to関係を実装している。
(rule (?x next-to ?y in (?x ?y . ?u)))
(rule (?x next-to ?y in (?v . ?z))

(?x next-to ?y in ?z))

以下のクエリに対する答えはどのようになるだろうか。
(?x next-to ?y in (1 (2 3) 4))
(?x next-to 1 in (2 1 3 1))

練習問題 4.62: 練習問題 2.17の last-pair演算を実装する規則を
定義せよ。これは、空でないリストの最後の要素を含むリストを
返すものである。(last-pair (3) ?x), (last-pair (1 2 3) ?x),
(last-pair (2 ?x) (3))のようなクエリに対して、それらの規則
をチェックせよ。それらの規則は (last-pair ?x (3))のような
クエリに対して正しく動作するだろうか。

練習問題 4.63: 次のデータベース (創世記第 4章参照)は、Adaの
⼦孫から家系をたどり、Cainを経て Adamに⾄るまで遡っている。
(son Adam Cain)
(son Cain Enoch)
(son Enoch Irad)
(son Irad Mehujael)
(son Mehujael Methushael)
(son Methushael Lamech)
(wife Lamech Ada)
(son Ada Jabal)
(son Ada Jubal)

“もし S が f の息⼦であり、かつ、f が Gの息⼦ならば、S は G

の孫である”、“もしW がM の妻であり、かつ、S がW の息⼦な
らば、S は M の息⼦である”といった (聖書時代には現代よりも
当てはまりやすかったと思われる)規則を定式化せよ。これらによ
って、クエリシステムが Cainの孫、Lamechの息⼦、Methushael
の孫を求めることができるようになる (より複雑な関係を推論す
る規則については、練習問題 4.69参照)。

4.4.2 クエリシステムの仕組み
この先の4.4.4節では、クエリインタプリタの実装を⼿続きの集合という形で
⽰す予定です。この節では、低レベルな実装の詳細によらないシステムの全体
的な構造についての概要を⾒ていくことにします。インタプリタの実装につい
て記述することによって、インタプリタの限界や、クエリ⾔語の論理演算が数
学論理学と微妙に違う点について、理解できるようになります。
クエリ評価器がクエリをデータベース内の事実や規則に対してマッチさせ

るためには何らかの探索を⾏う必要があるということはおわかりかと思いま
す。探索を⾏う⽅法のひとつは、4.3節の amb評価器を使って、クエリシステ
ムを⾮決定性プログラムとして実装するというものです (練習問題 4.78参照)。
別のやり⽅としては、ストリームの助けを借りて探索を管理するというものも
あります。ここでの実装は、この⼆つ⽬のアプローチを取ることにします。
このクエリシステムは、パターンマッチング (pattern matching)と ユニフ

ィケーション (unification)という⼆つの中⼼的な演算をベースに構築します。
初めにパターンマッチングについて述べ、フレームストリームという形で構成

した情報とパターンマッチングを合わせて使うことによって単純クエリと複合
クエリの両⽅を実装できるということを説明します。その次に、規則を実装す
るために必要となるパターンマッチングの⼀般化であるユニフィケーションに
ついて検討します。最後に、4.1節で記述したインタプリタで evalが式を分類
するのと似たようなやり⽅で式を分類する⼿続きによって、クエリインタプリ
タ全体がうまく組み⽴てられるということを⽰します。

パターンマッチング
パターンマッチャ (pattern matcher)とは、あるデータが指定されたパターン
に適合するかどうかをテストするプログラムです。例えば、データリスト ((a
b) c (a b))は、パターン (?x c ?x)にマッチし、このときパターン変数?xは
(a b)に束縛されます。この同じデータリストは、パターン (?x ?y ?z)にもマ
ッチし、このとき?xと?zはどちらも (a b)に束縛され、?yは cに束縛されま
す。このデータリストは、パターン ((?x ?y) c (?x ?y))にもマッチし、この
とき?xは aに、?yは bに束縛されます。しかし、これはパターン (?x a ?y)
にはマッチしません。このパターンは⼆つ⽬の要素が記号 aであるリストを指
定しているからです。
クエリシステムの使うパターンマッチャは、⼊⼒としてパターン、データ、

フレーム (frame)を取ります。フレームは、さまざまなパターン変数に対する
束縛を規定するものです。パターンマッチャは、フレームにすでに存在する束
縛と⽭盾せずにデータがパターンにマッチするかどうかをチェックします。マ
ッチするのであれば、マッチによって決められる束縛がある場合はその束縛を
与えられたフレームに追加して、与えられたフレームを返します。マッチしな
ければ、マッチが失敗したことを伝えます。
例えば、空のフレームが与えられているとき、パターン (?x ?y ?x)を使っ

て (a b a)にマッチさせると、?xが aに、?yが bに束縛されることを規定す
るフレームを返します。同じパターン、同じデータと、?yが aに束縛されるこ
とを規定するフレームによってマッチを試みると失敗します。同じパターン、
同じデータと、?yが bに束縛され?xが未束縛であるフレームによってマッチ
を試みると、?xの aへの束縛を与えられたフレームに追加したものが返され
ます。
規則を含まない単純なクエリを処理するためには、パターンマッチャとい

う仕組みがあれば⼗分です。例えば、次のクエリを処理するには、
(job ?x (computer programmer))

データベース内のすべての表明を探索し、初期状態が空であるフレームについ
てパターンにマッチするものを選択します。⾒つかったそれぞれのマッチに対
して、マッチによって返されたフレームを使って、?xの値でパターンを具体化
します。

フレームのストリーム
パターンのフレームに対するテストは、ストリームを使うことによって組み⽴
てられています。ひとつのフレームが与えられ、マッチング処理はデータベー
スのエントリをひとつずつ⾒ていきます。マッチャは、それぞれのデータベー
スエントリについて、マッチが失敗したことを⽰す特別な記号かフレームに対
する拡張のどちらかを⽣成します。全データベースエントリについての結果は
ストリームに集められ、フィルタを通して失敗が取り除かれます。結果は、デ
ータベースの何らかの表明に対するマッチによって与えられたフレームを拡張
したフレームすべてのストリームとなります。67

私たちのシステムでは、図 4.4に⽰すように、クエリはフレームの⼊⼒スト
リームを取り、ストリームの各フレームに対して上記のマッチング処理を⾏い
ます。つまり、クエリは⼊⼒ストリームの各フレームに対して、データベース
内の表明へのマッチによるそのフレームへの拡張すべてからなる新しいストリ
ームを⽣成するということです。次に、これらのストリームがすべて組み合わ
され、ひとつの巨⼤なストリームが作られます。このストリームは⼊⼒ストリ
ームの各フレームに対するすべての可能な拡張を含んでいます。このストリー
ムがクエリの出⼒となります。
単純クエリに答えるには、ひとつの空フレームからなる⼊⼒ストリームと

ともにクエリを使います。結果となる出⼒ストリームは、空フレームに対する
すべての拡張 (つまり、クエリに対するすべての答え)を含んでいます。次に、
このフレームのストリームを使って、元のクエリパターンのコピーの変数をそ
れぞれのフレーム内の値によって具体化したもののストリームを⽣成します。
これが最終的に表⽰されるストリームとなります。

67マッチングは⼀般的にとても⾼コストなので、完全なマッチャをデータベースの全要
素に適⽤するのは避けたいところです。通常これは、⾼速で粗いマッチと最終マッチに
処理を分解することによって達成します。粗いマッチはデータベースをフィルタし、最
終マッチにかける候補の⼩さな集合を作ります。うまくやると、粗いマッチのための作
業の⼀部を、候補選択時ではなくデータベース構築時に⾏うようにデータベースを作る
ことができます。これをデータベースのインデクシング (indexing)と⾔います。データ
ベースインデクシングの設計に関しては膨⼤な技術の積み上げがあります。4.4.4節で記
述するここでの実装は、そのような最適化を単純にしたものを含んでいます。

input stream
of frames

output stream of frames,
filtered and extendedquery

(job ?x ?y)

stream of assertions
from data base

図 4.4: フレームのストリームを処理するクエリ

複合クエリ
フレームのストリームという実装の真の美しさがわかるのは複合クエリを扱う
ときです。複合クエリの処理は、マッチが指定されたフレームと⽭盾しないよ
うに要求するというマッチャの能⼒を利⽤します。例えば、次のような⼆つの
クエリの andを扱うには、
(and (can-do-job ?x (computer programmer trainee))

(job ?person ?x))

(わかりやすく⾔うと、“プログラマ研修⽣の職務ができる⼈をすべて⾒つけ
よ”) まず次のパターンにマッチするすべてのエントリを探します。
(can-do-job ?x (computer programmer trainee))

すると、それぞれのフレームが?xに対する束縛を持つフレームのストリームが
できます。次に、ストリームの各フレームについて、与えられた?xに対する束
縛に⽭盾しない形で次のパターンにマッチするエントリをすべて探します。
(job ?person ?x)

そのようなマッチは、?xと?personに対する束縛を含むフレームをそれぞれ⽣
成します。⼆つのクエリの andは、図 4.5に⽰すように、部品となる⼆つのク
エリを直列に組み合わせたものとして⾒ることができます。最初のクエリフィ
ルタを通過したフレームが、⼆つ⽬のクエリによってフィルタされ、さらに拡
張されるということになります。

図 4.5: ⼆つのクエリの andによる組み合わせは、フレーム
のストリームに対する直列の演算によって⽣成される

図 4.6は、似たような考え⽅で、⼆つのクエリの orを、部品となる⼆つのク
エリを並列に組み合わせたものとして計算するという⼿法を⽰したものです。
⼊⼒となるフレームのストリームは、それぞれのクエリによって別々に拡張さ
れます。それから、結果となる⼆つのストリームがマージされ、最終的な出⼒
ストリームが⽣成されます。
これは⾼レベルの記述ですが、これだけ⾒ても複合クエリの処理速度が遅

くなりやすいということは明らかです。例えば、クエリはそれぞれの⼊⼒フレ
ームについて複数の出⼒ストリームを⽣成する可能性があり、andは前のクエ
リから⼊⼒フレームを受け取るため、andクエリは、最悪の場合、クエリ数に
対して指数的な数のマッチを実⾏する必要があります (練習問題 4.76参照)。68

単純クエリだけを扱うシステムはとても実⽤的なのですが、複合クエリを扱う
ことは⾮常に難しいのです。69

フレームのストリームという観点からは、クエリの notは、クエリが満た
されるフレームをすべて取り除くフィルタとしてふるまいます。例えば、次の
パターンが与えられたとき、
(not (job ?x (computer programmer)))

⼊⼒ストリームのそれぞれのフレームについて、(job ?x (computer programmer))
を満たす拡張フレームの⽣成を試みます。そして、そのような拡張が存在する

68しかし、このような指数的爆発は andクエリではあまりありません。追加条件は、⽣
成されるフレーム数を増やすより減らすことが多いからです。

69複合クエリの効果的な処理⽅法を扱うデータベース管理システムの⽂献は多数存在
します。

図 4.6: ⼆つのクエリの orによる組み合わせは、フレームの
ストリームに対して並列に演算を⾏い、その結果をマージ
することによって⽣成される

フレームすべてを⼊⼒ストリームから削除します。結果は、フレーム内の?xの
束縛が (job ?x (computer programmer))を満たさないフレームのみからなる
ストリームとなります。例えば、次のクエリの処理では、
(and (supervisor ?x ?y)

(not (job ?x (computer programmer))))

最初の節は?xと?yに対する束縛を持つフレームを⽣成します。次に not節は、
これらのフレームから、?xの束縛が?xがコンピュータプログラマであるとい
う制約を満たしているすべてのフレームを削除することによってフィルタリン
グします。70

lisp-value特殊形式は、フレームのストリームに対する同じようなフィル
タとして実装します。ストリームのそれぞれのフレームを使い、パターン内に

70フィルタによるこの notの実装と、数学論理学での通常の意味の notとの間には、微
妙な違いがあります。4.4.3節参照。

変数があれば具体化し、それから Lisp⼿続きを適⽤します。述語が失敗するフ
レームをすべて⼊⼒ストリームから削除します。

ユニフィケーション
クエリ⾔語の規則を扱うためには、結果部分が与えられたクエリパターンにマ
ッチする規則を⾒つけることができなければなりません。規則の結果は表明に
似ていますが、変数を持つことがあるというところが違います。そのため、パ
ターンマッチングを⼀般化したもの—ユニフィケーション (unification)と呼ば
れます—が必要となります。ユニフィケーションの場合は、“パターン”と “デ
ータ”の両⽅とも変数を持つ可能性があります。
ユニフィケーションを⾏うユニファイアは、⼆つのパターンを引数として

取ります。パターンはどちらも定数や変数を含みます。そして、⼆つのパター
ンが等しいものとなるような変数への値の割り当てが可能かどうかを決定しま
す。もし可能であれば、それらの束縛を含むフレームを返します。例えば、(?x
a ?y)と (?y ?z a)のユニフィケーションは、?x, ?y, ?zのすべてが aに束縛
されなければならないというフレームを規定します。⼀⽅、(?x ?y a)と (?x
b ?y)のユニフィケーションは失敗します。⼆つのパターンを等しいものとす
るような?yの値は存在しないからです (両⽅のパターンの⼆つ⽬の要素を等し
くするためには?yは bでなければなりません。しかし、三つ⽬の要素を等しく
するためには?yは aでなければなりません)。クエリシステムで使うユニファ
イアは、パターンマッチャ同様にフレームを⼊⼒として取り、そのフレームと
⽭盾しないようにユニフィケーションを実⾏します。
ユニフィケーションのアルゴリズムはクエリシステムの中で技術的に最も

難しい部分です。複雑なパターンの場合、ユニフィケーションの実⾏には推論
が必要となるように⾒えるかもしれません。例えば、(?x ?x)と ((a ?y c) (a
b ?z))のユニフィケーションを⾏うには、?xが (a b c)となり、?yが bとな
り、?zが cとなるということを推論しなければなりません。この処理は、パタ
ーンの部品同⼠の⽅程式を解くというように捉えることができます。⼀般的に、
これらは連⽴⽅程式であり、解くにはかなりの量の操作が必要となることもあ
ります。71 例えば、(?x ?x)と ((a ?y c) (a b ?z))のユニフィケーション
は、次の連⽴⽅程式を規定しているものとして考えることができます。
?x = (a ?y c)
?x = (a b ?z)

71⽚側パターンマッチングの場合は、パターン変数を含む等式はすべて明⽰的で、未知
のもの (パターン変数)についてはすでに解かれています。

これらの⽅程式は、
(a ?y c) = (a b ?z)

ということを含意しています。また、これは
a = a,

?y = b,
c = ?z,

ということを含意しています。よって、次のようになります。
?x = (a b c)

パターンマッチの場合は、マッチが成功するとパターン変数はすべて束縛され、
束縛される値は定数だけを含むようになっています。ここまでで⾒てきたユニ
フィケーションの例でも、すべてそのことが当てはまります。しかし⼀般的に
は、ユニフィケーションが成功した場合でも、変数の値は完全に決定されると
は限りません。変数は未束縛のまま残ることもあり、また変数を含む値に束縛
されることもあります。

(?x a)と ((b ?y) ?z)のユニフィケーションについて考えてみましょう。
?x = (b ?y)、a = ?zだということは推論できますが、?xと?yについてはそ
れ以上解くことはできません。このユニフィケーションは失敗はしません。?x
と?yに値を割り当てることによって⼆つのパターンを等しくすることは可能
だからです。このマッチは ?yの取る値についてはまったく制限していないた
め、?yに対する束縛は結果フレームに⼊れられません。しかし、このマッチ
は?xの値については制限を加えています。?yがどのような値であれ、?xは (b
?y)でなければなりません。そのため、?xの (b ?y)というパターンへの束縛
はフレームに⼊れられます。もし?yの値が (パターンマッチや、このフレーム
と⽭盾しないことが求められるユニフィケーションによって) 後で決定されて
フレームに追加されたら、その前に束縛された?xはこの値を参照することにな
ります。72

72ユニフィケーションについてのもうひとつの考え⽅は、⼆つの⼊⼒パターンの特殊
化となる最も⼀般的なパターンを⽣成するというものです。つまり、(?x a)と ((b ?y)
?z)のユニフィケーションは ((b ?y) a)となり、上で検討した (?x a ?y)と (?y ?z a)
のユニフィケーションは (a a a)となるということになります。ここでの実装では、ユ
ニフィケーションの結果はパターンとして考えるよりもフレームとして考えたほうが便
利です。

規則の適⽤
ユニフィケーションは、クエリシステム中の規則から推論を⾏う部分の鍵とな
っています。その実現⽅法を⾒るために、規則の適⽤が絡むクエリの処理につ
いて考えてみましょう。例えば、次のようなものです。
(lives-near ?x (Hacker Alyssa P))

このクエリを処理するためには、まず上で説明した通常のパターンマッチ⼿続
きを使って、このパターンにマッチする表明がデータベース内にあるかどうか
を調べます (この場合にはありません。データベースには、誰が誰の近所に住
んでいるかについての直接の表明が含まれていないからです)。次のステップ
として、クエリパターンと各規則の結果とのユニフィケーションを試みます。
すると、このパターンと、次の規則の結果をユニフィケーションすることによ
って、
(rule (lives-near ?person-1 ?person-2)

(and (address ?person-1 (?town . ?rest-1))
(address ?person-2 (?town . ?rest-2))
(not (same ?person-1 ?person-2))))

?person-2が (Hacker Alyssa P)に束縛され、?xが?person-1に束縛される
(?person-1と同じ値を持つ)ことを規定するフレームが返されるということが
わかります。今度は、このフレームについて、この規則の本体によって与えら
れる複合クエリを評価します。マッチが成功すると、このフレームは?person-
1に対する束縛を与えることによって拡張され、結果として?xの値も決まりま
す。?xの値があれば、それを使って元のクエリパターンを具体化できます。
⼀般的に、パターン変数に対する束縛を規定するフレーム内にクエリパタ

ーンを位置づけようとするとき、クエリ評価器は以下の⼿法を使って規則を適
⽤します。

• クエリを規則の結論とユニフィケーションし、成功すれば元のフレーム
を拡張したものを作る。

• 拡張したフレームについて、規則の本体によって作られるクエリを評価
する。

これは、Lispの eval/apply評価器での⼿続き適⽤⼿法にそっくりです。
• ⼿続きの仮引数を引数に束縛し、⼿続きの元の環境を拡張したフレーム
を作る。

• 拡張した環境について、⼿続きの本体によって作られる式を評価する
⼆つの評価器が似ているということは意外ではないでしょう。⼿続き定義が
Lispにおける抽象化の⼿段であるのと同じように、規則定義はクエリ⾔語にお
ける抽象化の⼿段です。どちらの場合も、適切な束縛を作成し、それらの束縛に
ついて規則や⼿続きの本体を評価することによって抽象化を解きほぐします。

単純なクエリ
この章では、規則のない単純クエリを評価するやり⽅をすでに学んでいます。
規則の適⽤⽅法についてもここまでで学んだため、規則と表明の両⽅を使って
単純クエリを評価するやり⽅についても記述できるようになりました。
クエリパターンとフレームのストリームが与えられたとき、⼊⼒ストリー

ムの各フレームに対して、⼆つのストリームを⽣成します。
• (パターンマッチャを使って)データベース内のすべての表明に対してパ
ターンをマッチさせることによって得られる拡張フレームのストリーム

• (ユニファイアを使って)すべての可能な規則を適⽤することにより得ら
れる拡張したフレームのストリーム73

これら⼆つのストリームを連結すると、元のフレームと⽭盾しないように与え
られたパターンを満⾜するすべてのやり⽅からなるストリームができます。次
に、これらのストリーム (⼊⼒ストリームのフレームごとにひとつずつ)がすべ
て結合され、ひとつの⼤きなストリームになります。このストリームは、元の
⼊⼒ストリームのいずれかを拡張して与えられたパターンとのマッチを作るや
り⽅すべてからなるものです。

クエリ評価器とドライバループ
基層となるマッチング演算は複雑なものですが、このシステムは普通の⾔語の
評価器とほぼ同じように構成されます。マッチング演算を司る⼿続きは qeval
というもので、Lispの eval⼿続きと似たような役割を果たします。qevalは、
⼊⼒としてクエリとフレームのストリームを取ります。出⼒はフレームのスト
リームで、それぞれのフレームは⼊⼒フレームのいずれかを拡張する成功した

73ユニフィケーションはマッチングの⼀般化であるため、両⽅のストリームを⽣成す
るのにユニファイアを使うことによってシステムを単純にすることもできるところです。
しかし、簡単なケースを単純なマッチャで扱うことによって、(ちゃんとしたユニフィケ
ーションではない)マッチングがそれ⾃⾝役に⽴つということがわかります。

マッチに対応しています。これは図 4.4で図⽰した通りです。eval同様、qeval
はいろいろなタイプの式 (クエリ)を分類し、それぞれに対する適切な⼿続きに
ディスパッチします。それぞれの特殊形式 (and, or, not, lisp-value)ごとに
ひとつ⼿続きがあり、また単純クエリの⼿続きもあります。
ドライバループは、この章のほかの評価器の driver-loop⼿続きと似てい

て、端末からクエリを読み込みます。それぞれのクエリについて、ドライバル
ープはそのクエリと空フレームひとつだけからなるストリームを引数として
qevalを呼び出します。これによって、すべての可能なマッチ (空フレームに
対するすべての可能な拡張)のストリームが⽣成されます。結果となるストリ
ームのそれぞれのフレームについて、ドライバループはフレームにある変数の
値を使って元のクエリを具体化します。その後、この具体化されたクエリのス
トリームが表⽰されます。74

ドライバは、特別コマンド assert!についてもチェックします。これは、⼊
⼒がクエリではなく、データベースに追加する表明または規則であるというこ
とを⽰すものです。次のように使われます。
(assert! (job (Bitdiddle Ben)

(computer wizard)))
(assert! (rule (wheel ?person)

(and (supervisor ?middle-manager ?person)
(supervisor ?x ?middle-manager))))

4.4.3 論理プログラミングは数学論理学か
クエリ⾔語で使われる組み合わせ⼿段は、最初は数学論理学の and, or, not演
算と同じように⾒えるかもしれません。また実際、クエリ⾔語の規則の適⽤は、
正当な推論の⼿法によってなされます。75 しかし、このようにクエリ⾔語と数
学論理学を同⼀視するのは、実際には妥当ではありません。クエリ⾔語は、論

74ここでフレームの (リストではなく)ストリームを使うのは、規則を再帰的に適⽤す
るとひとつのクエリを満たす無限の数の値が⽣成されることがあるからです。ストリー
ムに組み込まれた遅延評価がここで重要になります。システムは、応答の数が有限か無
限かにかかわらず、応答が⽣成されるごとにそれらをひとつずつ表⽰していきます。

75ある特定の推論⼿法が正当であるということは⾃明な主張ではありません。そのた
めには、真の前提から始めると真の結論のみが導出されるということを証明する必要が
あります。規則適⽤によって表現される推論の⼿法は、Aが真でありかつ Aならば Bが
真であるなら B は真であると結論づけるという、モーダス・ポネンス (modus ponens)
というよく知られた推論⼿法です。

理的な命題を⼿続き的に解釈する制御構造 (control structure)を持っているた
めです。この制御構造を活⽤することはよくあります。例えば、プログラマの
上司すべてを探すには、次の⼆つの論理的に等価な形式のどちらかによってク
エリを作ることができます。
(and (job ?x (computer programmer)) (supervisor ?x ?y))

(and (supervisor ?x ?y) (job ?x (computer programmer)))

もし会社にプログラマより多くの上司が存在する (それが⼀般的なケースです)
のであれば、⼆つ⽬の形式よりも⼀つ⽬の形式を使うほうがよいということに
なります。というのは、andの⼀つ⽬の節によって⽣成される中間結果 (フレー
ム)それぞれについてデータベースを⾛査する必要があるからです。
論理プログラミングが⽬標としているのは、計算上の問題を、“何を”計算

するのかと “どのように”これを計算するのかという⼆つの別々の問題に分解
する技術をプログラマに提供するということです。これを達成するために、計
算したいものを何でも記述できる程度には強く、かつ制御可能な⼿続き的解釈
が存在する程度には弱くなるように、数学論理学の命題の部分集合を選びます。
ここでの意図は、論理プログラミング⾔語によって規定されるプログラムは、
コンピュータによって実⾏できる効率的なプログラムでなければならないとい
う側⾯があるということです。制御 (“どのように”計算するか)は、⾔語の評価
の順序を使うことによって達成されるものです。ですので、節の順序や、それ
ぞれの節の中での副⽬標の順序を調整して、計算が実効的かつ効率的に⾏われ
ると思われる順序で実⾏されるようにできる必要があります。それと同時に、
計算の結果 (“何を”計算するか)は、論理法則の単純な帰結として⾒ることが
できる必要があります。
ここでのクエリ⾔語は、ちょうどそのような、⼿続き的に解釈可能な数学

論理学の部分集合と⾒なすことができます。表明は、単純な事実 (アトミック
な命題)を表現します。規則は含意を表現しています。この含意とは、規則の
本体が成り⽴つ場合に規則の結論が成り⽴つというものです。規則には、⾃然
な⼿続き的解釈があります。それは、規則の結論を成り⽴たせるためには、規
則の本体を成り⽴たせよ、というものです。これによって、規則は計算を規定
することになります。しかし、規則は数学的論理の命題と⾒なすこともできま
す。完全に数学的論理の範囲内で操作を⾏うことでも同じ結果が得られるわけ
ですから、論理プログラムによって⾏われるどのような “推論”も正しいもの
だと⾔えるわけです。76

76この記述には、「論理プログラムによってなされる “推論”というとき、その計算は

505



無限ループ
論理プログラムを⼿続き的に解釈することによる結果のひとつとして、ある種
の問題を解くのにどうしようもないほど⾮効率的なプログラムを構築してしま
うということがあります。⾮効率性の極端なケースは、推論をする中でシステ
ムが無限ループにはまってしまうというものです。単純な例として、有名な夫
婦のデータベースを作ることを考えてみましょう。そのデータベースには、次
のものが含まれています。
(assert! (married Minnie Mickey))

ここで、以下の質問をすると、
(married Mickey ?who)

応答は得られません。というのは、「Aが Bと結婚している場合、Bは Aと結
婚している」ということをシステムは知らないからです。そのため、以下の規
則を表明します。
(assert! (rule (married ?x ?y) (married ?y ?x)))

そして、同じ質問をします。
(married Mickey ?who)

残念ながら、こうするとシステムは次のように無限ループに陥ってしまいます。
• システムは married規則が適⽤可能だということを発⾒します。これは、
規則の結論 (married ?x ?y)はクエリパターン (married Mickey ?who)
とのユニファイに成功し、?xが Mickey、?yが?whoに束縛されるフレー
ムを⽣成するということです。そこで、インタプリタはこのフレームの中

停⽌するものとする」ということに合意する、という条件をつける必要があります。残
念ながら、私たちのクエリ⾔語の実装では、この条件は成り⽴ちません (これは Prolog
や、ほかのほとんどの現在の論理プログラミング⾔語のプログラムについても同様です)。
これは、notと lisp-valueを使っていることによります。この先で記述するように、ク
エリ⾔語に実装された notは数学論理学の notと常に⼀致するとは限らず、また lisp-
valueは事情をさらに複雑にします。単純に notと lisp-valueをこの⾔語から取り除
き、プログラムは単純クエリ、and、orだけで書くということにすれば、数学論理学と⼀
致する⾔語を実装できるところです。しかし、こうするとこの⾔語の表現⼒は⼤きく制
限されます。論理プログラミングの研究における主要な課題のひとつは、表現⼒を過度
に犠牲にすることなく、数学論理学とより⼀致するようにする⽅法を⾒つけるというこ
とです。

で規則本体の (married ?y ?x)を評価します。つまり、(married ?who
Mickey)というクエリを処理することになります。

• データベース内には、ひとつの答えが直接的に表明として存在します。
(married Minnie Mickey)

• しかしまた、married規則も適⽤可能です。そのため、インタプリタは
再度規則本体を評価します。今回は (married Mickey ?who)と等価にな
ります。

これでシステムは無限ループになります。システムがループに陥る前に単純な
答えである (married Minnie Mickey)を⾒つけられるかどうかは、システム
がデータベース内の項⽬をチェックする順番の実装に依存します。これは、起
こりうるループの⾮常に単純な例です。相互に関連する規則の集まりの場合、
これよりもずっと予測が難しいループを招くことがあり、またループが現れる
かどうかは and内の節の順序 (練習問題 4.64参照)や、システムがクエリを処
理する順序に関する低レベルの実装に依存することもあります。77

notの問題
クエリシステムのもうひとつの厄介なところは、notに関するものです。4.4.1
節のデータベースがあるとき、次の⼆つのクエリについて考えてみましょう。
(and (supervisor ?x ?y)

(not (job ?x (computer programmer))))
(and (not (job ?x (computer programmer)))

(supervisor ?x ?y))

77これは論理の問題ではなく、インタプリタによって提供される、論理の⼿続き的解釈
の問題です。ここでループに陥らないインタプリタを書くことも⼀応は可能です。例え
ば、表明と規則から導出可能なすべての証明を、深さ優先探索ではなく幅優先探索によっ
て列挙するというということもできます。しかし、そのようなシステムでは、プログラム
中の推論の順番を利⽤することが難しくなります。そのようなプログラムで精緻な制御
構造を構築する試みが deKleer et al. 1977に記述されています。そのような深刻な制御
上の問題を引き起こさない別のテクニックとして、特定の種類のループに対する検知器
といった、特別な知識を組み込むというものがあります (練習問題 4.67)。しかし、推論
を実⾏する際にシステムが無限の経路をたどり続けることを確実に防ぐ⼀般的な設計⽅
法はありえません。適切に選んだ関数 f についての、“P (x)が真であることを⽰すため
には、P (f(x))が真であることを⽰せ”という形の悪魔の規則を想像してみてください。

これらの⼆つのクエリは同じ結果を⽣成しません。⼀つ⽬のクエリは、ま
ずデータベース中の (supervisor ?x ?y) にマッチするすべてのエントリを
探し、その次に、返されたフレームの中から ?x の値が (job ?x (computer
programmer))を満たすものを削除します。⼆つ⽬のクエリは、まず⼊⼒フレ
ームをフィルタし、その中から (job ?x (computer programmer))を満たすも
のを削除します。唯⼀の⼊⼒フレームは空フレームなので、データベースから
(job ?x (computer programmer))を満たすパターンがあるかどうかチェック
します。普通はこの形のエントリが存在するので、not節は空フレームを取り
除き、空のフレームストリームを返します。その結果、複合クエリ全体は空ス
トリームを返すことになります。
問題は、notの実装が、実際は変数の値に対するフィルタとして働くように

作られているということです。not節が (上の例での?xのように)束縛されて
いないままの変数のあるフレームとともに処理されると、システムは期待しな
い結果を返します。同じような問題は、lisp-valueを使う際にも起こります。
Lispの述語は、未束縛の引数があると動作しません。練習問題 4.77を参照して
ください。
クエリ⾔語の notと数学論理学の notの違いは、これよりもはるかに深刻

なところにもあります。論理学では、命題 “not P”を、P は真ではないという
意味に解釈します。しかし、クエリシステムでは、“not P”は P がデータベー
ス内の知識から推論不可能であるという意味になります。例えば、4.4.1節の社
員情報データベースがあるとき、システムはどんな not命題でも、例えば Ben
Bitdiddleは野球ファンではない、外で⾬は降っていない、2 + 2 は 4ではない、
といったものでも、喜んで推論してくれるでしょう。78

練習問題 4.64: Louis Reasonerは誤って outranked-by規則 (4.4.1
節)をデータベースから削除してしまった。彼がこのことに気づい
たとき、すぐにまたその規則を組み込んだ。困ったことに、彼は規
則を少し変更してしまい、次のように⼊⼒した。
(rule (outranked-by ?staff-person ?boss)

(or (supervisor ?staff-person ?boss)

78(not (baseball-fan (Bitdiddle Ben)))というクエリについて考えてみましょう。
システムは、データベースに (baseball-fan (Bitdiddle Ben))がないということを発
⾒し、そのため空フレームはパターンを満たさないことになり、最初のフレームのスト
リームから取り除かれません。したがって、クエリの結果はその空フレームとなり、それ
が⼊⼒クエリを具体化するために使われて、(not (baseball-fan (Bitdiddle Ben)))
が⽣成されます。

(and (outranked-by ?middle-manager ?boss)
(supervisor ?staff-person

?middle-manager))))

Louisがこの情報をシステムに⼊⼒してすぐ DeWitt Aullがやっ
てきて、Ben Bitdiddleよりも地位が上の⼈には誰がいるかを調べ
ようとした。彼は次のクエリを⼊⼒した。
(outranked-by (Bitdiddle Ben) ?who)

応答の後、システムは無限ループに陥った。理由を説明せよ。
練習問題 4.65: 組織内での昇進の⽇を待ち望んでいる Cy D. Fect
は、(4.4.1節の規則 wheelを使って) 組織の重要⼈物をすべて検索
するクエリを⼊⼒してみた。
(wheel ?who)

彼はシステムからの応答に驚いた。
;;; Query results:
(wheel (Warbucks Oliver))
(wheel (Bitdiddle Ben))
(wheel (Warbucks Oliver))
(wheel (Warbucks Oliver))
(wheel (Warbucks Oliver))

Oliver Warbucksはなぜ四回表⽰されたのだろうか。
練習問題 4.66: Benは、会社についての統計データを出せるよう
にクエリシステムを⼀般化する作業をしているところだ。例えば、
以下のように⼊⼒することで、すべてのコンピュータプログラマ
の給料の合計を求めることができるようになる。
(sum ?amount (and (job ?x (computer programmer))

(salary ?x ?amount)))

⼀般的に、Benの新しいシステムは次の形式の式を受け⼊れる。
(accumulation-function ⟨variable⟩ ⟨query pattern⟩)

ここで、accumulation-functionは sum, average, maximumのよう
なものだ。Benは、この実装は朝飯前だろうと考えた。クエリパ
ターンを単純に qevalに与えればいい。そうするとフレームのス
トリームができる。それから、このストリームに、各フレームか
ら指定された変数の値を取り出す操作をマップし、その結果のス
トリームを集積関数に渡す。Benが実装を終えて動作確認をしよ
うとしていたところ、Cyがやってきた。Cyは相変わらず練習問
題 4.65の wheelクエリの結果に悩んでいた。Cyが Benにシステ
ムの応答を⾒せると、Benは “あぁ、ダメだ。これじゃこの単純な
集積計画はうまくいかないな”とうめいた。
Benは何に気づいたのだろうか。状況打開に使える⽅法の概要を
述べよ。
練習問題 4.67: 本⽂や練習問題 4.64で⽰したような種類の単純な
ループを防ぐために、クエリシステムにループ検知器を組み込む
⽅法を考えよ。考え⽅の概要は、システムが現在の推論のつなが
りについて何らかの形で履歴を管理するようにして、すでに実⾏
中のクエリの処理を始めないようにするというものである。この
履歴にはどのような種類の情報 (パターンとフレーム)が含まれる
か、またどのようにチェックを⾏うかを説明せよ (4.4.4節でクエ
リシステムの実装の細部を学んでから、システムを修正してルー
プ検知器を組み⼊れてみるのもいいだろう)。
練習問題 4.68: 与えられたリストと同じ要素を逆順で持つリスト
を返す、練習問題 2.18の reverse演算を実装する規則を定義せよ
(ヒント：append-to-formを使う)。その規則は (reverse (1 2 3)
?x)と (reverse ?x (1 2 3))の両⽅に答えることができるだろ
うか。
練習問題 4.69: 練習問題 4.63で作ったデータベースと規則をもと
に、孫 (grandson)の関係に “great”を追加する規則を考えよ。これ
によって、システムは Iradが Adamの great-grandson(ひ孫)であ
ることや、Jabalと JubalがAdamの great-great-great-great-great-
grandsons(ひひひひひ孫)であることが推論できるようになる (ヒ
ント：Iradについての事実を、例えば ((great grandson) Adam
Irad)と表現する。また、リストの末尾が単語 grandsonであるか
どうか決定する規則を書く。これを使って、?relが grandsonで終
わるリストであるとき、((great . ?rel) ?x ?y)という関係を導

出できる規則を表現する)。これらの規則を、((great grandson)
?g ?ggs)や (?relationship Adam Irad)のようなクエリによっ
て確認せよ。

4.4.4 クエリシステムの実装
4.4.2節では、クエリシステムの動作について述べた。ここではシステムの完全
な実装を紹介し、細部を埋めていきます。

4.4.4.1 ドライバループと具体化
クエリシステムのドライバループは繰り返し⼊⼒式を読み込みます。読み込ん
だ式がデータベースに追加するべき規則または表明であれば、その情報を追加
します。それ以外の場合は、式はクエリと⾒なされます。ドライバは、ひとつ
の空フレームからなる初期フレームストリームとともに、このクエリを評価器
qevalに渡します。評価の結果は、データベースに⾒つかる変数の値でクエリ
を満たすことによって⽣成されるフレームストリームです。これらのフレーム
は、元のクエリのコピーの変数をフレームストリームによって与えられる値に
よって具体化したものからなる新しいストリームを作るのに使われます。この
最後のストリームが端末に表⽰されます。
(define input-prompt ";;; Query input:")
(define output-prompt ";;; Query results:")

(define (query-driver-loop)
(prompt-for-input input-prompt)
(let ((q (query-syntax-process (read))))

(cond ((assertion-to-be-added? q)
(add-rule-or-assertion! (add-assertion-body q))
(newline)
(display "Assertion added to data base.")
(query-driver-loop))

(else
(newline)
(display output-prompt)
(display-stream
(stream-map

(lambda (frame)
(instantiate
q
frame
(lambda (v f)

(contract-question-mark v))))
(qeval q (singleton-stream '()))))

(query-driver-loop)))))

この章のほかの評価器と同じように、ここではクエリ⾔語の式に抽象構⽂を
使っています。式の構⽂の実装は、述語 assertion-to-be-added?やセレクタ
add-assertion-bodyも含め、4.4.4.7節にあります。add-rule-or-assertion!
は4.4.4.5節で定義されます。
⼊⼒式の処理を始める前に、ドライバループは⼊⼒式をより効率的に処理

できるような形に構⽂的に変換します。この変換にはパターン変数の表現の変
更も伴います。クエリが具体化されるときには、未束縛のままの変数があれば、
表⽰前に⼊⼒時の表現に再度変換されます。これらの変換は、query-syntax-
process、contract-question-markという⼆つの⼿続きによって⾏われます
(4.4.4.7節)。
式を具体化するときは、式をコピーし、中の変数をすべて与えられたフレ

ーム内の値で置き換えます。値⾃⾝も変数を含んでいる可能性があるため具体
化します (例えば、式の中の?xがユニフィケーションの結果?yに束縛され、?y
⾃⾝は 5に束縛されている場合)。変数が具体化できない場合に取るべき⾏動
は、instantiate⼿続きの引数として渡されます。
(define (instantiate exp frame unbound-var-handler)

(define (copy exp)
(cond ((var? exp)

(let ((binding (binding-in-frame exp frame)))
(if binding

(copy (binding-value binding))
(unbound-var-handler exp frame))))

((pair? exp)
(cons (copy (car exp)) (copy (cdr exp))))

(else exp)))
(copy exp))

束縛を操作する⼿続きは4.4.4.8節で定義されます。

4.4.4.2 評価器
query-driver-loopにより呼ばれる qeval⼿続きは、クエリシステムの基本的
な評価器です。⼊⼒としてクエリとフレームのストリームを取り、拡張したフ
レームのストリームを返します。特殊形式は、getと putを使ったデータ主導
のディスパッチによって識別します。これは、第 2章でジェネリック演算を実
装したときと同じです。
特殊形式と識別されない任意のクエリは単純クエリと⾒なされ、simple-

queryによって処理されます。
(define (qeval query frame-stream)

(let ((qproc (get (type query) 'qeval)))
(if qproc

(qproc (contents query) frame-stream)
(simple-query query frame-stream))))

typeと contentsは4.4.4.7節で定義されるもので、特殊形式の抽象構⽂を実装
します。

単純クエリ
simple-query⼿続きは単純クエリを扱います。引数として単純クエリ (パタ
ーン)とフレームのストリームを取り、それぞれのフレームをクエリのデータ
ベースとのマッチすべてによって拡張することで得られるストリームを返し
ます。
(define (simple-query query-pattern frame-stream)

(stream-flatmap
(lambda (frame)

(stream-append-delayed
(find-assertions query-pattern frame)
(delay (apply-rules query-pattern frame))))

frame-stream))

⼊⼒ストリームの各フレームについて、find-assertions(4.4.4.3節)を使って
パターンをデータベースの表明すべてとマッチさせ、拡張フレームのストリー
ムを⽣成します。そして、apply-rules(4.4.4.4節)を使って可能な規則をすべ
て適⽤し、また別の拡張フレームのストリームを⽣成します。これら⼆つのス
トリームは (stream-append-delayed(4.4.4.6節)を使って) 組み合わされ、結

果として与えられたパターンが元のフレームと⽭盾しない形で満たされるすべ
ての⽅法からなるストリームができます (練習問題 4.71参照)。個々の⼊⼒フレ
ームについてのこのストリームは stream-flatmap (4.4.4.6節)によって組み合
わされ、元の⼊⼒ストリームの任意のフレームを拡張して与えられたパターン
とのマッチを作るすべての⽅法の⼤きなストリームがひとつできます。

複合クエリ
andクエリは、図 4.5で図⽰したように、conjoin⼿続きによって扱われます。
conjoinは、⼊⼒として連⾔肢 (conjunct)とフレームのストリームを取り、拡
張されたフレームのストリームを返します。まず、conjoinはフレームのスト
リームを処理して、連⾔ (conjunction)の最初のクエリを満たすフレームの可
能な拡張をすべて探します。次に、これを新しいフレームのストリームとして、
クエリの残り部分に対して再帰的に conjoinを適⽤します。
(define (conjoin conjuncts frame-stream)

(if (empty-conjunction? conjuncts)
frame-stream
(conjoin (rest-conjuncts conjuncts)

(qeval (first-conjunct conjuncts) frame-stream))))

以下の式は、
(put 'and 'qeval conjoin)

qevalが and形式を⾒つけたときに conjoinにディスパッチするよう設定し
ます。
図 4.6に図⽰したように、orクエリも同じように扱います。orのそれぞれ

の選⾔肢 (disjunct) に対する出⼒ストリームは別々に計算され、4.4.4.6 節の
interleave-delayed⼿続きによって結合されます (練習問題 4.71、練習問題
4.72参照)。
(define (disjoin disjuncts frame-stream)

(if (empty-disjunction? disjuncts)
the-empty-stream
(interleave-delayed
(qeval (first-disjunct disjuncts) frame-stream)
(delay (disjoin (rest-disjuncts disjuncts) frame-stream)))))

(put 'or 'qeval disjoin)

連⾔肢構⽂と選⾔肢構⽂の述語とセレクタは4.4.4.7節で⽰します。

フィルタ
notは、4.4.2節で概説した⽅法で扱います。否定対象となるクエリを満たすよ
うに⼊⼒ストリームの各フレームを拡張することを試み、拡張できない場合の
み与えられたフレームを出⼒ストリームに⼊れます。
(define (negate operands frame-stream)

(stream-flatmap
(lambda (frame)

(if (stream-null?
(qeval (negated-query operands)

(singleton-stream frame)))
(singleton-stream frame)
the-empty-stream))

frame-stream))
(put 'not 'qeval negate)

lisp-valueは notに似たフィルタです。ストリームの各フレームを使ってパ
ターン内の変数を具体化し、指定された述語を適⽤し、述語が偽を返したフレ
ームを⼊⼒ストリームから取り除きます。未束縛のパターン変数がある場合は
結果はエラーとなります。
(define (lisp-value call frame-stream)

(stream-flatmap
(lambda (frame)

(if (execute
(instantiate
call
frame
(lambda (v f)

(error "Unknown pat var: LISP-VALUE" v))))
(singleton-stream frame)
the-empty-stream))

frame-stream))
(put 'lisp-value 'qeval lisp-value)

executeは述語を引数に適⽤するものですが、適⽤する⼿続きを得るためには
述語式を evalする必要があります。しかし、引数を評価してはいけません。引
数はすでに実際の引数であり、(Lispで)評価すると引数を⽣成する式ではない
からです。executeの実装には基層となる Lispシステムの evalと applyが使
われていることに注意してください。
(define (execute exp)

(apply (eval (predicate exp) user-initial-environment)
(args exp)))

特殊形式 always-trueは、常に満たされているクエリを提供します。always-
trueは中⾝ (通常は空)を無視し、⼊⼒ストリームのすべてのフレームをその
まま渡します。always-trueは、rule-bodyセレクタ (4.4.4.7節)が本体なしで
定義されている規則 (つまり、結論部分が常に満たされる規則)に対して本体を
提供するために使われます。
(define (always-true ignore frame-stream) frame-stream)
(put 'always-true 'qeval always-true)

notと lisp-valueの構⽂を定義するセレクタは4.4.4.7節で⽰します。

4.4.4.3 パターンマッチングによる表明の検索
find-assertionsは、simple-query (4.4.4.2節)に呼ばれるもので、⼊⼒とし
てパターンとフレームを取ります。返すのはフレームのストリームで、各フレ
ームは与えられたパターンに対するデータベースのマッチによって与えられ
たフレームを拡張したものです。この⼿続きは、パターンとフレームに対す
るマッチを確認しなければならないデータベース内のすべての表明のストリ
ームを得るために、fetch-assertions (4.4.4.5節)を使います。ここで fetch-
assertionsを⾏うのは、マッチが成功する候補のプールの中から多くのエン
トリを削除してくれる簡単なテストが適⽤できることがよくあるからです。こ
こで fetch-assertionsを省いてデータベース内のすべての表明のストリーム
をチェックしたとしてもシステムは依然として動作するでしょうが、マッチャ
の呼び出し回数がずっと多くなるので、計算は⾮効率的になるでしょう。
(define (find-assertions pattern frame)

(stream-flatmap
(lambda (datum) (check-an-assertion datum pattern frame))
(fetch-assertions pattern frame)))

check-an-assertionは引数としてパターン、データオブジェクト (表明)、フ
レームを取り、拡張されたフレームを含む 1要素のストリームか、マッチに失
敗した場合は the-empty-streamを返します。
(define (check-an-assertion assertion query-pat query-frame)

(let ((match-result
(pattern-match query-pat assertion query-frame)))

(if (eq? match-result 'failed)
the-empty-stream
(singleton-stream match-result))))

基本的なパターンマッチャは、記号 failed、または与えられたフレームを拡張
したものを返します。マッチャの基本的な考え⽅は、データに対して要素ごと
にパターンをチェックし、パターン変数に対する束縛を集積していくというも
のです。もしパターンとデータオブジェクトが同じなら、マッチは成功し、そ
こまでで集積された束縛のフレームを返します。同じでなければ、パターンが
変数の場合は、フレームにすでに存在する束縛と⽭盾しない範囲で、変数をデ
ータに束縛して現在のフレームを拡張します。パターンとデータが両⽅ともペ
アである場合は、(再帰的に)パターンの carをデータの carに対してマッチさ
せ、フレームを⽣成します。次に、このフレームの中でパターンの cdrをデー
タの cdrに対してマッチさせます。これらのどれにも当てはまらない場合、マ
ッチは失敗し、記号 failedを返します。
(define (pattern-match pat dat frame)

(cond ((eq? frame 'failed) 'failed)
((equal? pat dat) frame)
((var? pat) (extend-if-consistent pat dat frame))
((and (pair? pat) (pair? dat))
(pattern-match
(cdr pat)
(cdr dat)
(pattern-match (car pat) (car dat) frame)))

(else 'failed)))

以下は、フレームにすでにある束縛と⽭盾しない範囲で新しい束縛を追加して
フレームを拡張する⼿続きです。
(define (extend-if-consistent var dat frame)

(let ((binding (binding-in-frame var frame)))

(if binding
(pattern-match (binding-value binding) dat frame)
(extend var dat frame))))

対象となる変数についての束縛がフレーム内になければ、変数とデータの束縛
を追加します。束縛がある場合は、そのフレームの中で、データをフレーム内
の変数の値に対してマッチさせます。格納された値が定数しか持たないのであ
れば (extend-if-consistentによって格納されたのであれば必ずそうなって
いるはずです)、マッチは単純に格納された値と新しい値が同じかどうかテス
トします。同じであれば、変更されていないフレームを返します。同じでなけ
れば、失敗を表すものを返します。しかし実際には、値がユニフィケーション
の間に格納されたものである場合、格納された値がパターン変数を含んでいる
こともあります (4.4.4.4節参照)。格納されたパターンに対する新しいデータ
の再帰的なマッチは、このパターンに含まれる変数に対する束縛を追加したり
チェックしたりすることになります。例えば、?xが (f ?y)に束縛されていて、
?yが未束縛であるフレームがあるとして、このフレームを?xの (f b)への束
縛によって拡張したいとします。この場合、?xを検索し、それが (f ?y)に束
縛されているということを発⾒します。その結果、提案された新しい値である
(f b)に対して、同じフレームで (f ?y)をマッチさせることになります。最
終的に、このマッチは?yの bへの束縛を追加することによってこのフレームを
拡張します。?xは (f ?y)に束縛されたままです。格納されている束縛を修正
することはありません。また、与えられた変数に対して複数の束縛を格納する
こともありません。

extend-if-consistentによって使われる、束縛を操作する⼿続きは4.4.4.8
節で定義されています。

ドット末尾を持つパターン
パターンがドットとそれに続くパターン変数を含む場合、練習問題 2.20で述べ
たドット末尾記法から予想できるように、そのパターン変数はデータリストの
(次の要素ではなく) 残りにマッチします。ここまでで実装したパターンマッチ
ャはドットを探すことはしませんが、期待通りのふるまいをします。これは、
query-driver-loopで使っている Lispの基本⼿続き readが、クエリを読み込
んでそれをリスト構造として表現する際に、ドットを特別扱いするためです。

readがドットを⾒つけると、次の項⽬をリストの次の要素にする (consの
carにして、その cdrがリストの残りになる)のではなく、次の項⽬をリスト
構造の cdrにします。例えば、パターン (computer ?type)に対する readに

518



よって⽣成されるリスト構造は式 (cons 'computer (cons '?type '()))を評
価することによって構築されるものになり、パターン (computer . ?type)の
場合は式 (cons 'computer '?type)を評価することによって構築されるもの
になります。
そのため、pattern-matchがデータリストとドットを持つパターンの car

と cdrを再帰的に⽐較していくと、最終的にはドットの次の変数 (パターンの
cdr)がデータリストの部分リストに対してマッチされることになり、変数はそ
のリストに対して束縛されることになります。例えば、パターン (computer .
?type)を (programmer computer trainee)にマッチさせると、?typeをリス
ト (programmer trainee)にマッチさせることになります。

4.4.4.4 規則とユニフィケーション
apply-rulesは、find-assertions(4.4.4.3節)の規則版です。⼊⼒としてパタ
ーンとフレームを取り、データベースの規則を適⽤することによって拡張した
フレームのストリームを作ります。stream-flatmapは、(4.4.4.5節の fetch-
rulesによって選択された)適⽤可能かもしれない規則のストリームに対して
apply-a-ruleをマップし、返されるフレームのストリームを結合します。
(define (apply-rules pattern frame)

(stream-flatmap (lambda (rule)
(apply-a-rule rule pattern frame))

(fetch-rules pattern frame)))

apply-a-ruleは4.4.2節で概説した⼿法によって規則を適⽤します。まず、与
えられたフレーム内で規則の結論とパターンのユニフィケーションを⾏うこと
によって引数のフレームを拡張します。それが成功すると、この新しいフレー
ム内で規則の本体を評価します。
しかし、これらのことを⾏う前に、プログラムは規則内のすべての変数にユ

ニークな名前をつけ直します。その理由は、別々の規則適⽤の変数が互いに混
同されないようにするためです。例えば、⼆つの規則が両⽅とも?xという名前
の変数を使っている場合、規則適⽤時にどちらも?xに対する束縛をフレームに
追加するかもしれません。これら⼆つの?xは互いに関係がないので、この⼆つ
の束縛に⽭盾があってはならないなどと考えるようになってはいけません。変
数名を変えるのではなく、より巧妙な環境構造を考えることもできるところか
もしれません。しかし、変数名を変えるというここで選んだやり⽅は、最も効
率のよいものではないにしても、最も素直なやり⽅です (練習問題 4.79参照)。
apply-a-rule⼿続きは以下のようになります。

(define (apply-a-rule rule query-pattern query-frame)
(let ((clean-rule (rename-variables-in rule)))

(let ((unify-result (unify-match query-pattern
(conclusion clean-rule)
query-frame)))

(if (eq? unify-result 'failed)
the-empty-stream
(qeval (rule-body clean-rule)

(singleton-stream unify-result))))))

セレクタ rule-body と conclusion は規則の部品を取り出すもので、4.4.4.7
節で定義されます。
ユニークな変数名を作る際には、規則適⽤ごとにユニークな識別⼦ (例えば

数字)を関連づけ、その識別⼦と元の変数名を組み合わせることにします。例え
ば、規則適⽤の識別⼦が 7なら、規則内の?xをすべて?x-7に、?yをすべて?y-
7に変更することになります (make-new-variableと new-rule-application-
idは 4.4.4.7節の構⽂⼿続きに含まれています)。
(define (rename-variables-in rule)

(let ((rule-application-id (new-rule-application-id)))
(define (tree-walk exp)

(cond ((var? exp)
(make-new-variable exp rule-application-id))

((pair? exp)
(cons (tree-walk (car exp))

(tree-walk (cdr exp))))
(else exp)))

(tree-walk rule)))

ユニフィケーションのアルゴリズムは、⼊⼒として⼆つのパターンとフレーム
を取り、拡張されたフレームまたは記号 failedを返す⼿続きとして実装され
ます。ユニファイアはパターンマッチャに似ていますが、対称的だというと
ころが違います— 変数はマッチのどちら側に現れることも可能です。unify-
matchは基本的には pattern-matchと同じですが、マッチの右側のオブジェク
トが変数である場合を扱う (下で “***”をつけた)追加コードがあるという違
いがあります。
(define (unify-match p1 p2 frame)

(cond ((eq? frame 'failed) 'failed)
((equal? p1 p2) frame)
((var? p1) (extend-if-possible p1 p2 frame))
((var? p2) (extend-if-possible p2 p1 frame)) ; ***
((and (pair? p1) (pair? p2))
(unify-match (cdr p1)

(cdr p2)
(unify-match (car p1)

(car p2)
frame)))

(else 'failed)))

ユニフィケーションでは、⽚側パターンマッチと同じように、既存の束縛と⽭
盾しないときだけフレーム拡張の提案を受け⼊れたいところです。ユニフィケ
ーションで使われる⼿続き extend-if-possibleは、パターンマッチで使われ
る extend-if-consistentと同じものですが、下記のプログラムで “***”をつ
けた⼆箇所の特別なチェックの部分が違います。⼀つ⽬のチェックでは、マッ
チさせようとしている変数が未束縛で、それに対してマッチさせようとしてい
る値がそれ⾃⾝ (別の)変数であるとき、その値が束縛されているかを確認する
必要があり、束縛されていればその値をマッチさせる必要があります。マッチ
の両側がどちらも未束縛であれば、どちらかをもう⼀⽅に束縛します。
⼆つ⽬のチェックは、ある変数を、その変数を含むパターンに束縛しよう

としている場合を扱います。このような状況は、ひとつの変数が両⽅のパター
ン内で繰り返し出てくる場合、いつでも起こる可能性があります。例えば、?x
と?yのどちらも未束縛のフレームで、(?x ?x)と (?y <?yを含む式 >)という
⼆つのパターンのユニフィケーションを⾏う場合について考えてみましょう。
まず、?xは?yにマッチされ、?xから?yへの束縛が作られます。次に、同じ?x
が?yを含む与えられた式にマッチされます。?xはすでに?yに対して束縛され
ているため、結果として ?yをその式にマッチさせることになります。⼆つの
パターンを同じものにするようなパターン変数の値の集合を探すものとしてユ
ニファイアを考えるなら、これらのパターンは ?yが?yを含む式に等しくなる
ような?yを探せという指⽰を意味することになります。そのような⽅程式を
解く⼀般的な⼿法はないため、そのような束縛は却下します。これらのケース
は、述語 depends-on?によって識別されます。79 ⼀⽅、変数をそれ⾃⾝に束縛
しようとするのは拒否したくないところです。例えば、(?x ?x)と (?y ?y)の

79⼀般的に、?yと、?yを含む式とのユニフィケーションを⾏う場合、式?y = <?yを
含む式 >の不動点を⾒つけることができる必要があります。時には、解に⾒えるような

ユニフィケーションについて考えてみましょう。?xを⼆回⽬に?yに束縛しよ
うとするとき、?y(?xに格納された値)と ?y(?xの新しい値)をマッチさせるこ
とになります。これは unify-matchの equal?節で処理します。
(define (extend-if-possible var val frame)

(let ((binding (binding-in-frame var frame)))
(cond (binding

(unify-match (binding-value binding) val frame))
((var? val) ; ***
(let ((binding (binding-in-frame val frame)))

(if binding
(unify-match
var (binding-value binding) frame)

(extend var val frame))))
((depends-on? val var frame) ; ***
'failed)

式を構⽂的に作ることができることもあります。例えば、?y = (f ?y)は、不動点 (f (f
(f . . . )))を持つように⾒えます。これは、式 (f ?y)から始めて、?yを繰り返し (f
?y)に置き換えていくことによって⽣成できます。残念ながら、そのような式がすべて意
味のある不動点を持つわけではありません。ここでの問題は、数学で無限級数を操作す
る問題と似ています。例えば、私たちは 2が式 y = 1 + y/2の解であることを知ってい
ます。式 1 + y/2から始めて、yを繰り返し 1 + y/2に置き換えていくと、以下のように
なります。

2 = y = 1 +
y

2
= 1 +

1

2

(
1 +

y

2

)
= 1 +

1

2
+

y

4
= . . . ,

これは以下の式になります。

2 = 1 +
1

2
+

1

4
+

1

8
+ . . . .

しかし、式 y = 1 + 2yの解が-1であるというところから始めると、
−1 = y = 1 + 2y = 1 + 2(1 + 2y) = 1 + 2 + 4y = . . . ,

これは以下の式になります。
−1 = 1 + 2 + 4 + 8 + . . . .

これらの⼆つの式を導くのに使った形式的操作は同じものなのに、⼀つ⽬の結果は無限
級数についての正当な主張になっていて、⼆つ⽬はそうなっていません。同じように、こ
こでのユニフィケーションの結果についても、構⽂的に構築した勝⼿な式によって推論
しようとすると、誤りを招く可能性があります。

(else (extend var val frame)))))

depends-on?は、パターン変数の値として提案された式がその変数に依存する
かどうかテストする述語です。これは現在のフレームに対して⾏う必要があり
ます。式は、すでにテスト対象の変数に依存する値を持つ変数を含んでいるか
もしれないからです。depends-on?の構造は⽊に対する簡単な再帰探索となり、
その中で必要に応じて変数の値を置き換えていくことになります。
(define (depends-on? exp var frame)

(define (tree-walk e)
(cond ((var? e)

(if (equal? var e)
true
(let ((b (binding-in-frame e frame)))

(if b
(tree-walk (binding-value b))
false))))

((pair? e)
(or (tree-walk (car e))

(tree-walk (cdr e))))
(else false)))

(tree-walk exp))

4.4.4.5 データベースの保守
論理プログラミング⾔語の設計で重要な問題のひとつは、与えられたパターン
をチェックする際に、無関係なデータベースのエントリを検査することをでき
るだけ避けられるようにするということです。ここでのシステムでは、すべて
の表明をひとつの⼤きなストリームに格納するのに加えて、carが定数記号で
あるすべての表明を、その記号によって索引づけされたテーブルの中に別々の
ストリームとして格納します。パターンにマッチするかもしれない表明を取り
出すときは、まずパターンの carが定数記号かどうかチェックします。定数記
号であれば、格納された表明の中で同じ carを持つものを返します (それをマ
ッチャでテストすることになります)。パターンの carが定数記号でなければ、
格納された表明をすべて返します。もっと賢いやり⽅としては、フレーム内の
情報を活⽤したり、パターンの carが定数記号でない場合にも最適化を試みた
りといったことも考えられます。索引づけの基準 (carを使うこと、定数記号の

523



場合のみを扱うこと)はプログラム内に組み込まないようにして、代わりにそ
の基準を具体化した述語とセレクタを呼び出すことにします。
(define THE-ASSERTIONS the-empty-stream)
(define (fetch-assertions pattern frame)

(if (use-index? pattern)
(get-indexed-assertions pattern)
(get-all-assertions)))

(define (get-all-assertions) THE-ASSERTIONS)
(define (get-indexed-assertions pattern)

(get-stream (index-key-of pattern) 'assertion-stream))

get-streamはテーブル内のストリームを検索します。何も格納されていなけ
れば空ストリームを返します。
(define (get-stream key1 key2)

(let ((s (get key1 key2)))
(if s s the-empty-stream)))

規則も同じように規則の結論の carによって格納されます。しかし、規則の結
論は任意のパターンであるため変数を含むことがあるという点が表明とは違い
ます。carが定数記号であるパターンは、結論が同じ carを持つ規則に加えて、
結論が変数で始まる規則にもマッチする可能性があります。そのため、carが
定数記号であるパターンにマッチするかもしれない規則を取り出す場合は、結
論がそのパターンと同じ carを持つ規則に加えて、結論が変数で始まる規則も
すべて取り出します。このために、結論が変数で始まるすべての規則をテーブ
ル内の独⽴したストリームに格納します。索引には記号?を使います。
(define THE-RULES the-empty-stream)
(define (fetch-rules pattern frame)

(if (use-index? pattern)
(get-indexed-rules pattern)
(get-all-rules)))

(define (get-all-rules) THE-RULES)
(define (get-indexed-rules pattern)

(stream-append
(get-stream (index-key-of pattern) 'rule-stream)
(get-stream '? 'rule-stream)))

add-rule-or-assertion!は、query-driver-loopが表明と規則をデータベー
スに追加するのに使います。それぞれの項⽬は、条件を満たせば索引に格納さ
れ、それからデータベース内のすべての表明と規則のストリームに格納され
ます。
(define (add-rule-or-assertion! assertion)

(if (rule? assertion)
(add-rule! assertion)
(add-assertion! assertion)))

(define (add-assertion! assertion)
(store-assertion-in-index assertion)
(let ((old-assertions THE-ASSERTIONS))

(set! THE-ASSERTIONS
(cons-stream assertion old-assertions))

'ok))
(define (add-rule! rule)

(store-rule-in-index rule)
(let ((old-rules THE-RULES))

(set! THE-RULES (cons-stream rule old-rules))
'ok))

実際に表明または規則を格納する際には、索引づけが可能かどうかチェックし
ます。可能であれば、適切なストリームに格納します。
(define (store-assertion-in-index assertion)

(if (indexable? assertion)
(let ((key (index-key-of assertion)))

(let ((current-assertion-stream
(get-stream key 'assertion-stream)))

(put key
'assertion-stream
(cons-stream
assertion
current-assertion-stream))))))

(define (store-rule-in-index rule)
(let ((pattern (conclusion rule)))

(if (indexable? pattern)
(let ((key (index-key-of pattern)))

(let ((current-rule-stream
(get-stream key 'rule-stream)))

(put key
'rule-stream
(cons-stream rule

current-rule-stream)))))))

以下の三つの⼿続きは、データベースの索引の使い⽅を定義するものです。パ
ターン (表明または規則の結論)は、先頭が変数または定数記号であればテーブ
ルに格納されます。
(define (indexable? pat)

(or (constant-symbol? (car pat))
(var? (car pat))))

パターンが格納されるテーブルのキーは、?(パターンの先頭が変数である場合)
またはパターン先頭の定数記号です。
(define (index-key-of pat)

(let ((key (car pat)))
(if (var? key) '? key)))

パターンが定数記号で始まるのであれば、パターンにマッチするかもしれない
項⽬を取得するために索引が使われます。
(define (use-index? pat) (constant-symbol? (car pat)))

練習問題 4.70: ⼿続き add-assertion!と add-rule!内の letに
よる束縛の⽬的は何か。以下に⽰す add-assertion!の実装はど
こが間違っているだろうか。ヒント：3.5.2節での (define ones
(cons-stream 1 ones)) という 1の無限ストリームの定義を思い
出そう。
(define (add-assertion! assertion)

(store-assertion-in-index assertion)
(set! THE-ASSERTIONS

(cons-stream assertion THE-ASSERTIONS))
'ok)

4.4.4.6 ストリーム演算
クエリシステムでは、第 3章で紹介していないいくつかのストリーム演算を使
っています。

stream-append-delayed, interleave-delayedは stream-append、interleave(3.5.3
節)と似ていますが、(3.5.4節の integralのように)遅延化された引数を取る
というところが違います。これはループを先送りすることがあります (練習問
題 4.71参照)。
(define (stream-append-delayed s1 delayed-s2)

(if (stream-null? s1)
(force delayed-s2)
(cons-stream
(stream-car s1)
(stream-append-delayed
(stream-cdr s1)
delayed-s2))))

(define (interleave-delayed s1 delayed-s2)
(if (stream-null? s1)

(force delayed-s2)
(cons-stream
(stream-car s1)
(interleave-delayed
(force delayed-s2)
(delay (stream-cdr s1))))))

stream-flatmapは、フレームのストリームに⼿続きをマップして結果となる
複数のフレームのストリームを結合するためにクエリ評価器全体を通して使わ
れるものですが、これは2.2.3節で通常のリストに対して導⼊した flatmap⼿
続きのストリーム版です。しかし、stream-flatmapは通常の flatmapと違っ
て、単純にストリームを⾜していくのではなく、結果を互い違いに挟み込むと
いうプロセスによって集積します。

(練習問題 4.72と練習問題 4.73参照)

(define (stream-flatmap proc s)
(flatten-stream (stream-map proc s)))

(define (flatten-stream stream)

(if (stream-null? stream)
the-empty-stream
(interleave-delayed
(stream-car stream)
(delay (flatten-stream (stream-cdr stream))))))

評価器は、次の単純な⼿続きによって要素ひとつからなるストリームを作るこ
ともあります。
(define (singleton-stream x)

(cons-stream x the-empty-stream))

4.4.4.7 クエリ構⽂⼿続き
typeと contentsは qeval(4.4.4.2節)によって使われるもので、ある特殊形式
がその carの記号によって識別されるということを規定します。これらはエラ
ーメッセージ以外は2.4.2節の type-tagと contents⼿続きと同じものです。
(define (type exp)

(if (pair? exp)
(car exp)
(error "Unknown expression TYPE" exp)))

(define (contents exp)
(if (pair? exp)

(cdr exp)
(error "Unknown expression CONTENTS" exp)))

以下の⼆つの⼿続きは4.4.4.1節の query-driver-loopによって使われるもの
で、規則と表明が (assert! <rule-or-assertion>)の形式の式によってデー
タベースに追加されるということを規定します。
(define (assertion-to-be-added? exp)

(eq? (type exp) 'assert!))
(define (add-assertion-body exp) (car (contents exp)))

以下は特殊形式 and, or, not, lisp-valueのための構⽂定義です (4.4.4.2節)。
(define (empty-conjunction? exps) (null? exps))
(define (first-conjunct exps) (car exps))
(define (rest-conjuncts exps) (cdr exps))

(define (empty-disjunction? exps) (null? exps))
(define (first-disjunct exps) (car exps))
(define (rest-disjuncts exps) (cdr exps))
(define (negated-query exps) (car exps))
(define (predicate exps) (car exps))
(define (args exps) (cdr exps))

以下の三つの⼿続きは規則の構⽂を定義します。
(define (rule? statement)

(tagged-list? statement 'rule))
(define (conclusion rule) (cadr rule))
(define (rule-body rule)

(if (null? (cddr rule)) '(always-true) (caddr rule)))

query-driver-loop(4.4.4.1節)は query-syntax-processを呼び、?symbolと
いう形を持つ式のパターン変数を (? symbol)という内部形式に変形します。
つまり、(job ?x ?y)のようなパターンは、実際には内部ではシステムによっ
て (job (? x) (? y))と表現されているということです。これによってクエ
リ処理の効率が向上します。こうすると、式がパターン変数かどうかシステム
がチェックする際に、式の carが記号?かどうかチェックすればよく、記号か
ら⽂字を抽出しないですむことになるからです。構⽂変形は以下の⼿続きによ
って実⾏されます。80

(define (query-syntax-process exp)
(map-over-symbols expand-question-mark exp))

(define (map-over-symbols proc exp)
(cond ((pair? exp)

(cons (map-over-symbols proc (car exp))
(map-over-symbols proc (cdr exp))))

80ほとんどの Lisp システムでは、ユーザーがリーダーマクロ⽂字 (reader macro
character)を定義することによって、通常の read⼿続きにそのような変形を⾏わせるこ
とができるようになっています。クォート式はすでにこのような扱いになっています。読
み込み⼿続きは、評価器が式を⾒る前に、'expressionを⾃動的に (quote expression)
に変形します。同じようにして、?expressionが (? expression)と変形されるように
することもできるところです。しかし、ここではわかりやすさのために変形⼿続きを明
⽰的に⼊れました。
expand-question-markと contract-question-markは、名前に stringを含むいくつ

かの⼿続きを使っています。これらは Schemeの基本⼿続きです。

((symbol? exp) (proc exp))
(else exp)))

(define (expand-question-mark symbol)
(let ((chars (symbol->string symbol)))

(if (string=? (substring chars 0 1) "?")
(list '?

(string->symbol
(substring chars 1 (string-length chars))))

symbol)))

変数をこのように変形すると、パターン内の変数は? で始まるリストになり
ます。定数記号 (データベースの索引づけのために識別する必要があります。
4.4.4.5節)はただの記号です。
(define (var? exp) (tagged-list? exp '?))
(define (constant-symbol? exp) (symbol? exp))

規則適⽤時には、ユニークな変数が構築されます (4.4.4.4節)。規則適⽤のユニ
ーク識別⼦は数値で、規則適⽤のたびにインクリメントされます。
(define rule-counter 0)
(define (new-rule-application-id)

(set! rule-counter (+ 1 rule-counter))
rule-counter)

(define (make-new-variable var rule-application-id)
(cons '? (cons rule-application-id (cdr var))))

query-driver-loopが答えを表⽰するためにクエリを具体化する際には、以下
の⼿続きを使って未束縛のパターン変数をすべて表⽰⽤の形式に戻します。
(define (contract-question-mark variable)

(string->symbol
(string-append "?"

(if (number? (cadr variable))
(string-append (symbol->string (caddr variable))

"-"
(number->string (cadr variable)))

(symbol->string (cadr variable))))))

4.4.4.8 フレームと束縛
フレームは束縛のリストとして表現されます。束縛は変数と値のペアです。
(define (make-binding variable value)

(cons variable value))
(define (binding-variable binding) (car binding))
(define (binding-value binding) (cdr binding))
(define (binding-in-frame variable frame)

(assoc variable frame))
(define (extend variable value frame)

(cons (make-binding variable value) frame))

練習問題 4.71:
Louis Reasonerは、simple-queryと disjoinの⼿続き (4.4.4.2節)
がなぜ以下のような定義ではなく、明⽰的な delay演算を⽤いて
実装されているのだろうかと不思議に思っている。
(define (simple-query query-pattern frame-stream)

(stream-flatmap
(lambda (frame)

(stream-append
(find-assertions query-pattern frame)
(apply-rules query-pattern frame)))

frame-stream))
(define (disjoin disjuncts frame-stream)

(if (empty-disjunction? disjuncts)
the-empty-stream
(interleave
(qeval (first-disjunct disjuncts)

frame-stream)
(disjoin (rest-disjuncts disjuncts)

frame-stream))))

これらの簡単な定義を使った場合に望ましくないふるまいを引き
起こすクエリの例を挙げよ。
練習問題 4.72: disjoinと stream-flatmapは、なぜそれらを単純
につなげていくのではなく、ストリームを互い違いに挟み込むよ

うにしているのか。なぜ相互配置のほうがよりうまくいくのかを
⽰す例を挙げよ。(ヒント：3.5.3節ではなぜ interleaveを使った
のだろうか)

練習問題 4.73: なぜ flatten-streamは明⽰的に delayを使って
いるのだろうか。以下のように定義するとどのような問題がある
のだろうか。
(define (flatten-stream stream)

(if (stream-null? stream)
the-empty-stream
(interleave
(stream-car stream)
(flatten-stream (stream-cdr stream)))))

練習問題 4.74: Alyssa P. Hackerは、negate, lisp-value, find-
assertionsの中で簡単なほうの stream-flatmapを使うことを提
案した。彼⼥は、これらの場合にフレームのストリームにマップ
される⼿続きが作るのは常に空ストリームか単⼀要素のストリー
ムであるため、これらのストリームを組み合わせるのに互い違い
に挟み込んでいく必要はないと気がついていた。

a. Alyssaのプログラムに⽋けている式を埋めよ。
(define (simple-stream-flatmap proc s)

(simple-flatten (stream-map proc s)))
(define (simple-flatten stream)

(stream-map ⟨??⟩
(stream-filter ⟨??⟩ stream)))

b. このような変更を加えると、クエリシステムのふるまいは変
わるだろうか。

練習問題 4.75: クエリ⾔語に uniqueと呼ばれる新しい特殊形式を
実装せよ。uniqueは、指定されたクエリを満たす項⽬がデータベ
ース中にちょうどひとつだけある場合に成功する。例えば、
(unique (job ?x (computer wizard)))

この式は、⼀項⽬からなる次のストリームを表⽰する。

(unique (job (Bitdiddle Ben) (computer wizard)))

これは、Benが唯⼀のコンピュータウィザードであるからだ。また、
(unique (job ?x (computer programmer)))

この式は空ストリームを表⽰する。コンピュータプログラマは⼆
⼈以上いるからだ。さらに、
(and (job ?x ?j) (unique (job ?anyone ?j)))

この式は、それをしている⼈が⼀⼈しかいない職位と、それをし
ている⼈たちをすべて表⽰する。
uniqueの実装は⼆つの部分に分かれる。⼀つ⽬の部分はこの特殊
形式を扱う⼿続きを書くためのもので、⼆つ⽬の部分は qevalが
この⼿続きにディスパッチするようにするためのものだ。qevalは
ディスパッチをデータ主導によって⾏うため、⼆つ⽬の部分は簡
単だ。⼿続きの名前が uniquely-assertedであれば、次のように
書けばいい。
(put 'unique 'qeval uniquely-asserted)

これで、qevalは型 (car)が記号 uniqueであるすべてのクエリを
この⼿続きにディスパッチする。本当の問題は⼿続き uniquely-
assertedを書くことである。これは⼊⼒として uniqueクエリの
contents(cdr)とフレームのストリームを受け取る。次に、ストリ
ームのそれぞれのフレームに対し、qevalを使って与えられたク
エリを満たすフレームの拡張すべてのストリームを得る。ちょう
どひとつの項⽬を持っているストリーム以外は取り除く。残った
ストリームが返され、それらがひとつの巨⼤なストリームに集積
されて、uniqueクエリの結果になる。これは特殊形式 notの実装
に似ている。
ちょうど⼀⼈の部下を持つ⼈をすべて列挙するクエリを作り、実
装をテストせよ。
練習問題 4.76: ここでのシステムでは、andをクエリの直列の組み
合わせ (図 4.5)として実装している。この実装はエレガントだが、
andの⼆つ⽬のクエリを処理する際に⼀つ⽬のクエリが⽣成した
フレームそれぞれについてデータベースを⾛査しなければならな

いため⾮効率的だ。データベースが n個の要素を持っているとし
て、典型的なクエリが nに⽐例した数 (n/k個とする)の出⼒フレ
ームを⽣成するとすると、⼀つ⽬のクエリによって⽣成されたフ
レームそれぞれに対してデータベースを⾛査するためには n2/k回
パターンマッチャを呼び出す必要がある。別のアプローチとして
は、andの⼆つの節を別々に処理し、互いに⽭盾しない出⼒フレ
ームのペアをすべて探すというものもある。それぞれのクエリが
n/k 個の出⼒フレームを⽣成するとすると、⽭盾テストは n2/k2

回実⾏する必要がある。つまり、必要なマッチの回数が現在の⼿
法よりも k倍少ないということになる。
この戦略による andの実装を考えよ。実装する⼿続きは⼊⼒とし
て⼆つのフレームを取り、それらのフレームの中の束縛が互いに
⽭盾しないか確認し、⽭盾しないのであれば束縛の⼆つの集合を
マージしたフレームを⽣成する。この演算はユニフィケーション
に似ている。
練習問題 4.77: 4.4.3節では、変数が束縛されていないフレームに
notや lisp-valueといったフィルタリング演算を適⽤すると、ク
エリ⾔語が “間違った” 答えを返すことがあるということを学ん
だ。この⽋陥を直す⽅法を考えよ。ひとつの考え⽅としては、フ
ィルタリングを “遅延化された”やり⽅で実⾏するというものがあ
る。そのためには、フレームにフィルタリングを⾏うという “約
束”を追加し、その約束は演算が可能となる⼗分な数の変数が束縛
されたら実⾏することにする。フィルタリングの実⾏は、ほかの
すべての演算が実⾏されてからにすることも考えられる。しかし、
効率のためには、⽣成する中間フレームを減らせるようにフィル
タリングをできるだけ早く実⾏したい。
練習問題 4.78: クエリ⾔語をストリーム処理として実装するので
はなく、4.3節の評価器によって実装する⾮決定性プログラムとし
て再設計せよ。このアプローチでは、それぞれのクエリは (すべて
の回答のストリームではなく)ひとつの回答を⽣成し、ユーザーは
try-againを⼊⼒することによって回答の続きを⾒ることができ
ることになる。この節で構築した仕組みの多くは⾮決定性探索と
バックトラックに含まれていたということに気づくはずだ。しか
し、それと同時に、新しいクエリ⾔語のふるまいがここまでで実
装したクエリ⾔語のふるまいと微妙な違いがあるということにも

気づくだろう。その違いを⽰すような例を⾒つけることができる
だろうか。
練習問題 4.79: 4.1節で Lisp評価器を実装したとき、局所環境を
使って別々の⼿続きの引数同⼠で名前が衝突しないようにする⽅
法を学んだ。例えば、以下を評価する際に、
(define (square x) (* x x))
(define (sum-of-squares x y)

(+ (square x) (square y)))
(sum-of-squares 3 4)

squareの xと sum-of-squaresの xが混同されることはない。そ
れぞれの⼿続きの本体は、局所変数の束縛を持たせるために特別
に構築した環境の中で評価されるからだ。クエリシステムでは、規
則適⽤時の名前衝突を避けるために違う戦略を使った。規則適⽤
のたびに、ユニークであることが保証された新しい名前に変数の
名前を変えている。Lisp評価器で同じような戦略を使うとすると、
局所環境をなくし、単純に⼿続き適⽤のたびに⼿続きの本体の中
の変数をリネームするということになるだろう。
クエリ⾔語で、変数のリネームではなく、環境を使う規則適⽤の
⼿法を実装せよ。その環境構造の上に、ブロック構造⼿続きの規
則版のような、⼤規模システムを扱うためのクエリ⾔語の構築物
を作ることはできるだろうか。これらの問題と、問題解決の⼿法
として⽂脈の中で推論を⾏うという問題 (例えば、“もし P が真で
あると仮定すると、Aと Bが推論できる”)とを関連づけることは
できるか。(この問題は⾃由回答だ。うまい解答ができれば博⼠号
が取れるだろう。)

5
レジスタマシンによる計算

私の⽬的は、天の機械は神聖なものや⽣きたものなどではなく、⼀
種の時計仕掛けであるということを⽰すことです (時計が魂を持
つと信じる⼈は、その栄光を時計の作り⼿ではなく、その仕掛け
に帰しているのです)。すべての多様な動きは⾮常に単純で物質的
な⼒によって引き起こされていて、それは時計のすべての動きが
ひとつのおもりに引き起こされているのと同じことなのです。
—Johannes Kepler (Herwart von Hohenburgへの⼿紙, 1605)

この本は、プロセスについて考察し、Lispで書かれた⼿続きによってプロセス
を記述するというところから始めました。これらの⼿続きの意味を説明するた
めに、いくつかの評価モデルを使いました。第 1章の置換モデル、第 3章の環
境モデル、第 4章のメタ循環評価器です。特に、メタ循環評価器を詳しく⾒て
いくことで、Lisp的⾔語の解釈についての謎を解くことができました。しかし、
メタ循環評価器もひとつの重要な問題に答えていません。これは、この評価器
が Lispシステムの制御メカニズムを明らかにしていないことによります。例
えば、部分式の評価がどのようにしてその値を使う式に値を返すのか、あるい
は、ある種の再帰関数は反復プロセスを⽣成する (つまり、定数空間で評価さ

れる)のに他の再帰関数は再帰プロセスになってしまうのはなぜか、といった
ところを、この評価器は説明してくれません。これらの疑問に対する答えがな
いのは、メタ循環評価器がそれ⾃⾝ Lispプログラムであり、基層となる Lisp
システムの制御構造を引き継いでいるからです。Lisp評価器の制御構造につい
てより完全な記述をするためには、Lispそのものよりも基本的なレベルで⾒て
いく必要があります。
この章では、従来式コンピュータのステップごとの演算によってプロセス

を記述していきます。そのようなコンピュータ、別名レジスタマシン (register
machine)は、レジスタ (register)と呼ばれる⼀組の固定された記憶素⼦の中⾝
を操作する 命令 (instruction)を順に実⾏していきます。典型的なレジスタマ
シンの命令は、レジスタの中⾝に基本的な演算を適⽤し、その結果をほかのレ
ジスタに割り当てるというものです。レジスタマシンにより実⾏されるプロセ
スをこのように記述すると、従来式コンピュータの “機械語”に⾮常によく似
ているように⾒えます。しかし、ここではある特定のコンピュータの機械語を
扱うのではなく、いくつかの Lisp⼿続きについて調べ、それぞれの⼿続きを実
⾏するための個別のレジスタマシンを設計していくことにします。そのため、
課題に対する私たちのアプローチは、機械語のプログラマというよりも、ハー
ドウェア設計者としての視点によるものとなります。レジスタマシンを設計す
る際には、再帰のような重要なプログラム構造を実装する仕組みを開発してい
きます。また、レジスタマシンの設計を記述するための⾔語も導⼊します。5.2
節では、私たちが設計するマシンをこの⾔語による記述法を利⽤してシミュレ
ートする Lispプログラムを実装します。
私たちのレジスタマシンの基本演算は、ほとんどが⾮常に単純なものです。

例えば、ある演算は⼆つのレジスタから取ってきた値を⾜し、答えを出して別
のレジスタに格納します。そのような演算は簡単に記述したハードウェアで実
⾏できます。しかし、リスト構造を扱うには、複雑な記憶領域割り当て機構が
必要となる car, cdr, consといったメモリ演算も使います。5.3節では、より
初歩的な演算によってこれらを実装する⽅法を学びます。

5.4節では、簡単な⼿続きをレジスタマシンとして定式化する経験を積んだ
後に、4.1節のメタ循環評価器で説明したアルゴリズムを実⾏するマシンを設
計します。これは、評価器の制御メカニズムを明⽰するモデルの役割を果たし、
Schemeの式の解釈がどのように⾏われるかについての理解のギャップを埋め
てくれるでしょう。5.5節では、評価器レジスタマシンのレジスタと演算によ
って直接実⾏可能な命令列に Schemeプログラムを翻訳する単純なコンパイラ
について学びます。

5.1 レジスタマシンの設計
レジスタマシンを設計するためには、まずレジスタマシンのデータパス (data
path) (レジスタと演算)と、それらの演算を正しく並べるコントローラ (con-
troller)を設計する必要があります。簡単なレジスタマシンの設計について⽰
すために、⼆つの整数の最⼤公約数 (gcd)を求めるために使われるユークリッ
ドのアルゴリズムについて⾒ていきましょう。1.2.5節で学んだように、ユーク
リッドのアルゴリズムは反復プロセスによって実⾏可能です。プロセスは以下
の⼿続きによって規定されます。

(define (gcd a b)
(if (= b 0)

a
(gcd b (remainder a b))))

このアルゴリズムを実⾏するマシンは、aと bという⼆つの数値を管理する必
要があります。ですので、ここではそれらの数値が aと bという⼆つのレジス
タに格納されているということにします。必要となる基本演算は、レジスタ b
の値が 0かどうかを確認すること、レジスタ aの中⾝をレジスタ bの中⾝で
割った剰余を求めることの⼆つです。剰余を求める演算は複雑なプロセスです
が、当⾯は剰余を求める基本装置があると仮定します。gcdアルゴリズムのそ
れぞれのサイクルでは、レジスタ bの中⾝によってレジスタ aの中⾝を置き換
えることと、aの古い中⾝を bの古い中⾝で割った剰余によって bの中⾝を置
き換えることが必要になります。これら⼆つの置換を同時に⾏うことができれ
ば都合がいいのですが、私たちのレジスタマシンのモデルでは、各ステップで
新しい値を割り当てることができるのはレジスタひとつだけだということにし
ます。⼆つの置換を⾏うには、私たちのマシンは三つ⽬の “⼀時的”レジスタを
使います。これは tという名前にします。(まず、剰余を tに⼊れます。次に、
bの中⾝を aに⼊れます。最後に、tに格納されている剰余を bに⼊れます。)
このマシンで必要となるレジスタと演算は、図 5.1のようなデータパス図に

よって⽰すことができます。この図では、レジスタ (a, b, t)は⻑⽅形で表して
います。レジスタに対する値の割り当ては、それぞれ途中に Xのある⽮印によ
って⽰しています。⽮印はデータの代⼊元からレジスタという⽅向になってい
ます。この Xは、押すと代⼊元から指定されたレジスタに値を “流す”ボタン
と考えることができます。それぞれのボタンの横にあるラベルは、それぞれの
ボタンを呼ぶときに使う名前です。この名前は任意で、覚えやすいように名前

a b

t

rem

a ← b

t ← r

b ← t

0

=

図 5.1: gcdマシンのデータパス

をつけることができます (例えば、a<-bはボタンを押すとレジスタ bの中⾝を
aに割り当てるということを⽰しています)。レジスタに対する代⼊元は (a<-b
の代⼊のように)別のレジスタであったり、(t<-rの代⼊のように)演算の結果
であったり、定数 (変更できない組み込みの値で、データパス図では定数を持
つ三⾓形で表します)であることもあります。
定数やレジスタの中⾝から値を計算する演算は、データパス図では演算の

名前を持つ台形で表します。例えば、図 5.1で remというラベルをつけた箱は、
その箱に接続された aレジスタと bレジスタの中⾝の剰余を求める演算を表し
ます。⼊⼒レジスタや⼊⼒定数から箱に向かって (ボタンのない)⽮印があり、
また演算の出⼒値からレジスタに対しても⽮印が出ています。テストは、テス
トの名前を持つ円で表します。例えば、私たちの gcdマシンには、レジスタ b
の中⾝がゼロかどうかをテストする演算があります。テストにも⼊⼒レジスタ
や⼊⼒定数からの⽮印はありますが、出⼒の⽮印はありません。その値は、デ
ータパスではなくコントローラによって使われます。全体として、データパス
図はこのマシンに必要なレジスタと演算と、またそれらをどう接続するべきか
を⽰しています。⽮印を配線と考え、Xボタンをスイッチだと考えると、デー
タパス図は電⼦部品によって構築できる機械の配線図にそっくりです。
データパスが実際に gcdを求めるには、ボタンが正しい順序で押されなけ

ればいけません。この順序については、図 5.2に⽰すようなコントローラ図に
よって記述します。コントローラ図の要素は、データパスの部品がどう操作さ
れるべきかを⽰しています。コントローラ図の⻑⽅形の箱は押されるべきデー
タパスのボタンを特定し、⽮印はあるステップから次のステップに移る順序を

= t ← r

a ← b

b ← t

図 5.2: gcdマシンのコントローラ

⽰しています。図の中の菱形は判定を表しています。菱形に対応するデータパ
スのテストの値によって、⼆つの⽮印のどちらかに進むことになります。コン
トローラは、物理的なアナロジーを使って解釈できます。コントローラ図をビ
ー⽟が転がっている迷路として考えてみましょう。ビー⽟が箱に⼊ると、箱に
つけられた名前のデータパスのボタンを押します。ビー⽟が (b = 0のテスト
のような)判定ノードに⼊ると、そのノードの⽰すテストの結果によって決ま
るルートに進みます。データパスとコントローラは、⼆つ合わせて gcdを計算
するマシンを完全に記述していることになります。レジスタ aと bに値をセッ
トしてから、コントローラ (転がるビー⽟)を startという印の場所から始め
ます。コントローラが doneに到達するときには、gcdの値がレジスタ aの中に
⼊っています。

練習問題 5.1: 以下の⼿続きによって規定されるような、反復アル
ゴリズムによって階乗を計算するレジスタマシンを設計せよ。こ
のマシンに対するデータパス図とコントローラの図を描け。
(define (factorial n)

(define (iter product counter)
(if (> counter n)

product
(iter (* counter product)

(+ counter 1))))

(iter 1 1))

5.1.1 レジスタマシンの記述⾔語
データパス図とコントローラ図は、gcdのような単純なマシンを表現するのに
は向いています。しかし、Lispインタプリタのような⼤きなマシンを記述しよ
うと思うと⼤変です。複雑なマシンを扱えるように、データパス図とコントロ
ーラ図のすべての情報をテキスト形式で表現する⾔語を作ることにします。ま
ずは、図を直接反映した表記から始めます。
マシンのデータパスは、レジスタと演算を記述することによって定義しま

す。レジスタを記述するには、名前をつけ、そのレジスタへの代⼊を制御する
ボタンを規定します。これらのボタンそれぞれにも名前をつけ、そのボタンの
制御によってレジスタに⼊るデータの⼊⼒元を規定します (⼊⼒元はレジスタ、
定数、演算です)。演算を記述するには、名前をつけ、その⼊⼒元 (レジスタま
たは定数)を規定します。
マシンのコントローラは、命令 (instruction)とラベル (label)の列として定

義します。ラベルは、列のエントリポイント (entry point)を⽰します。命令は
次のうちのいずれかです。

• データパスボタンの名前。その名前のボタンを押してレジスタに値を割
り当てる (これはコントローラ図の箱に対応する)。

• test(テスト)命令。規定されたテストを実⾏する。
• コントローララベルの指す場所への条件分岐 (branch命令)。条件分岐は
直前のテストの結果に基づいて⾏われる (テストと分岐は、⼆つ合わせ
てコントローラ図の菱形に対応する)。テストが偽であれば、コントロー
ラは命令列の次の命令に進む。真であれば、コントローラはそのラベル
の次の命令に進む。

• 無条件分岐 (goto命令)。実⾏を移すべきコントローララベルを指定する。
マシンはコントローラの命令列の先頭からスタートし、実⾏が列の終わりにた
どり着くと停⽌します。分岐が制御の流れを変えるとき以外は、命令は並べら
れている順に実⾏されます。

図 5.3: ↓ gcdマシンの仕様
(data-paths
(registers

((name a)
(buttons ((name a<-b) (source (register b)))))

((name b)
(buttons ((name b<-t) (source (register t)))))

((name t)
(buttons ((name t<-r) (source (operation rem))))))

(operations
((name rem) (inputs (register a) (register b)))
((name =) (inputs (register b) (constant 0)))))

(controller
test-b ; label

(test =) ; test
(branch (label gcd-done)) ; conditional branch
(t<-r) ; button push
(a<-b) ; button push
(b<-t) ; button push
(goto (label test-b)) ; unconditional branch

gcd-done) ; label

図 5.3は、この⽅法で記述した gcdマシンを⽰したものです。この例では、こ
の記述⽅法の⼀般性を垣間⾒ることしかできません。gcdマシンはとても単純
なケースだからです。このマシンでは、それぞれのレジスタはボタンをひとつ
しか持たず、それぞれのボタンとテストはコントローラに⼀回しか使われてい
ません。
残念ながら、このような記述は読むのが⼤変です。コントローラの命令を

理解するには何回もボタン名と命令名の定義に戻る必要があり、ボタンの動作
について理解するには演算名の定義を参照する必要がある場合もあります。そ
のため、データパスの記述とコントローラの記述の情報を組み合わせて⼀度に
⾒られるように表記法を変形します。
この形式による記述を作るには、適当に決めていたボタンの名前と演算の

名前を、ふるまいの定義に基づくものに変更します。つまり、(コントローラ
で)“ボタン t<-rを押せ”と⾔ったり、(データパスで)“ボタン t<-rは rem演算
の値をレジスタ tに代⼊する”、“rem演算の⼊⼒はレジスタ aとレジスタ bの
中⾝である”と⾔ったりする代わりに、今後は (コントローラで)“レジスタ aと
レジスタ bの中⾝に対する rem演算の値をレジスタ tに代⼊するボタンを押
せ”と⾔うことにします。同様に、(コントローラの中で) “=テストを実⾏せよ”

と⾔って、それとは別に (データパスの中で) “=テストはレジスタ bの中⾝と
定数 0に対して実⾏する”と⾔ったりする代わりに、今後は “=テストをレジス
タ bの中⾝と定数 0に対して実⾏せよ”と⾔うことにします。データパスの記
述は取り除き、コントローラの命令列だけが残ります。その結果、gcdマシン
は以下のような記述になります。
(controller
test-b

(test (op =) (reg b) (const 0))
(branch (label gcd-done))
(assign t (op rem) (reg a) (reg b))
(assign a (reg b))
(assign b (reg t))
(goto (label test-b))

gcd-done)

この形式による記述は、図 5.3で⽰したものよりも読みやすくなっていますが、
⽋点もあります。

• この形式は、⼤きなマシンにはより冗⻑になります。データパス要素が
コントローラの命令列に出てくるたびに、要素の完全な記述が繰り返さ
れることになるからです (これは gcdの例では問題になりません。演算
とボタンはそれぞれ⼀度しか使われないからです)。さらに、データパス
の記述を繰り返すことによって、マシンの実際のデータパス構造がわか
りにくくなっています。⼤きなマシンでは、レジスタ、演算、ボタンが
それぞれいくつあって、それらがどのように相互接続されているかすぐ
にはわかりません。

• マシンの定義に含まれるコントローラの命令は Lisp式に似ているため、
どんな Lisp式でもいいわけではないということを忘れてしまいがちで
す。コントローラの命令はマシンの演算として正当なものしか記述でき
ません。例えば、演算が直接扱うことができるのは定数とレジスタの中
⾝だけです。ほかの演算の結果を扱うことはできません。

これらの⽋点はありますが、この章を通して、このレジスタマシン⽤⾔語を使
うことにします。ここでは、データパスの要素や接続よりも、コントローラを
理解することに重点を置いているからです。しかし、実際にマシンを設計する
際にはデータパスの設計が⾮常に重要だということは意識しておく必要があり
ます。

 (controller
  gcd-loop
    (assign a (op read))
    (assign b (op read))
  test-b
    (test (op =) 
          (reg b) 
          (const 0))
    (branch (label gcd-done))
    (assign t 
            (op rem) 
            (reg a) 
            (reg b))
    (assign a (reg b))
    (assign b (reg t))
    (goto (label test-b))
  gcd-done
    (perform (op print) 
             (reg a))
    (goto (label gcd-loop)))

read

a b

t

rem

a ← b

t ← r

b ← t

0

=

print

a ← rd b ← rd

P

図 5.4: ⼊⼒を読み込み結果を表⽰する gcdマシン

練習問題 5.2: レジスタマシン⾔語を使って練習問題 5.1の反復階
乗マシンを記述せよ。

アクション
gcdマシンを修正して、gcdを求めたい数値の組を⼊⼒し、端末に答えが表⽰
されるようにしてみましょう。読み込みや表⽰の機能を持つマシンの作り⽅に
ついては考えずに、(Schemeで readや displayを使うときのように)それらは
基本演算として使えるようになっていると仮定します。1

readは、レジスタに格納できる値を返すという意味で、これまで使ってき
た演算に似ています。しかし、readはレジスタから⼊⼒を取ってくるわけでは
ありません。値は、私たちが設計しているマシンの外側で起きる何かによって

1この仮定はかなりの複雑さをごまかしています。Lispシステムでは、読み込みと表
⽰を機能させるために実装のかなりの部分が費やされるのが普通です。

決まります。ここでは、マシンの演算がそのようなふるまいを持つことを許す
ことにして、図や記述の中では値を計算するほかの演算と同じように readを
表記します。
⼀⽅、printはこれまで使ってきた演算と根本的に違います。printはレジ

スタに格納する値を⽣成しません。この演算は効果を持っていますが、設計し
ているマシンの⼀部に対するものではありません。この種の演算はアクション
(action)と呼ぶことにします。データパス図では、アクションは値を計算する
演算と同じように、つまりアクション名を中に含む台形として表現します。ア
クションの箱には任意の⼊⼒から⽮印が⼊っていきます。また、アクションに
はボタンも関連づけます。ボタンを押すとアクションが起こります。コントロ
ーラにアクションボタンを押させるには、perform(実⾏)という新しい種類の
命令を使います。つまり、レジスタ aの中⾝を表⽰するアクションは、コント
ローラの命令列の中では次の命令によって表現されます。
(perform (op print) (reg a))

新しい gcdマシンのデータパスとコントローラを図 5.4に⽰します。回答を表
⽰した後は、マシンを停⽌させるのではなく再開させ、数値のペアを読み込み、
それらの gcdを計算し、結果を表⽰することを繰り返すようにしています。こ
の構造は、第 4章のインタプリタで使ったドライバループに似ています。

5.1.2 マシン設計における抽象化
ここから先で、実際には⾮常に複雑な “基本”演算を含めたマシンを設計する
ということをよく⾏います。例えば、5.4節と5.5節では、Schemeの環境操作
を基本演算として扱います。このような抽象化を⾏うことによって、マシンの
⼀部の細部を無視して設計のほかの側⾯に集中できるという意味で、抽象化は
役に⽴ちます。しかし、多くの複雑な部分に蓋をしたからといって、マシンの
設計が⾮現実的だということではありません。複雑な “基本演算”は、より簡単
な演算に置き換えることがいつでもできます。

gcdマシンの場合を考えてみましょう。このマシンは、レジスタ aとレジス
タ bの中⾝の剰余を計算して結果をレジスタ tに割り当てる命令を持っていま
す。もし基本演算としての剰余演算を使わずに gcdマシンを構築したければ、
例えば引き算のようなより単純な演算によって剰余を計算する⽅法を規定する
必要があります。実際に、この⽅法で剰余を求める Schemeの⼿続きは次のよ
うに書くことができます。
(define (remainder n d)

(if (< n d)
n
(remainder (- n d) d)))

ですので、gcdマシンのデータパス内の剰余演算は、引き算演算と⽐較テ
ストで置き換えることができます。図 5.5に、そのように詳細化したマシンの
データパスとコントローラを⽰します。gcdコントローラ定義内の
(assign t (op rem) (reg a) (reg b))

という命令は、図 5.6に⽰すように、ループを含む命令列によって置き換える
ことができます。
図 5.6: ↓ 図 5.5の gcdマシンのコントローラの命令列

(controller test-b
(test (op =) (reg b) (const 0))
(branch (label gcd-done))
(assign t (reg a))

rem-loop
(test (op <) (reg t) (reg b))
(branch (label rem-done))
(assign t (op -) (reg t) (reg b))
(goto (label rem-loop))

rem-done
(assign a (reg b))
(assign b (reg t))
(goto (label test-b))

gcd-done)

練習問題 5.3: 1.1.7節に⽰したニュートン法によって平⽅根を求
めるマシンを設計せよ。
(define (sqrt x)

(define (good-enough? guess)
(< (abs (- (square guess) x)) 0.001))

(define (improve guess)
(average guess (/ x guess)))

(define (sqrt-iter guess)

図 5.5: 詳細化した gcdマシンのデータパスとコントローラ

(if (good-enough? guess)
guess
(sqrt-iter (improve guess))))

(sqrt-iter 1.0))

最初は good-enough?と improve演算が基本命令として使えると
仮定して設計せよ。次に、それらを算術演算によって展開する⽅
法を⽰せ。データパス図を描き、レジスタマシン⾔語でコントロ
ーラ定義を書くことによって、それぞれのバージョンの sqrtマシ
ンの設計を記述せよ。

5.1.3 サブルーチン
計算を⾏うマシンを設計する際には、部品を複製するのではなく、計算のいろ
いろな部分で共有できるようにしたいということがよくあります。⼆つの gcd
計算を持つマシンについて考えてみましょう。ひとつはレジスタ aとレジスタ
bの中⾝の gcdを求めるもので、もうひとつはレジスタ cとレジスタ dの gcd
を求めるものとします。この場合、まず基本演算としての gcd演算があると仮
定し、それから⼆つの gcdの実体をより基本的な演算を使って展開するという
ことが考えられます。図 5.7は、その結果としてできるマシンのデータパスの
gcd部分だけを⽰したものです。それらがマシンの残りの部分とどのように接
続されているかは⽰していません。この図には、マシンのコントローラ命令列
の中でそれらに対応する部分も⽰しています。
このマシンには剰余演算の箱が⼆つと等価テストの箱が⼆つあります。重

複する部品が、例えば剰余の箱のように複雑なものであれば、これはマシンを
構築するのに経済的な⽅法とはいえないでしょう。外側のマシンの計算に影響
しない形でどちらの gcdの計算にも同じ部品を使うようにすれば、データパス
部品の重複を避けることができます。もしコントローラが gcd-2に来たときに
レジスタ aとレジスタ bの値が必要ないのであれば (または、これらの値をほ
かのレジスタに退避できるのであれば)、⼀つ⽬の gcdを求めるときと同じく、
⼆つ⽬の gcdを求めるときにも、レジスタ cとレジスタ dを使う代わりにレジ
スタ aと bを使うようにマシンを変更できます。これをすると、図 5.8に⽰す
ようなコントローラ命令列が得られます。

gcd-1
 (test (op =) (reg b) (const 0))
 (branch (label after-gcd-1))
 (assign t (op rem) (reg a) (reg b))
 (assign a (reg b))
 (assign b (reg t))
 (goto (label gcd-1))
after-gcd-1

a b

t

rem

a ← b

t ← r

b ← t

0

=

gcd-2
 (test (op =) (reg d) (const 0))
 (branch (label after-gcd-2))
 (assign s (op rem) (reg c) (reg d))
 (assign c (reg d))
 (assign d (reg s))
 (goto (label gcd-2))
after-gcd-2

c d

s

rem

c ← d

s ← r

d ← s

0

=

図 5.7: ⼆つの gcdの計算を持つマシンのデータパスとコン
トローラ命令列の⼀部

図 5.8: ↓ 別々の⼆つの gcd計算に同じデータパスの部品を使うマ
シンのコントローラ命令列の⼀部
gcd-1
(test (op =) (reg b) (const 0))
(branch (label after-gcd-1))
(assign t (op rem) (reg a) (reg b))
(assign a (reg b))
(assign b (reg t))
(goto (label gcd-1))

after-gcd-1
. . .

gcd-2
(test (op =) (reg b) (const 0))

(branch (label after-gcd-2))
(assign t (op rem) (reg a) (reg b))
(assign a (reg b))
(assign b (reg t))
(goto (label gcd-2))

after-gcd-2

重複するデータパス部品は削除しました (これでデータパスはまた図 5.1のよ
うになります)、コントローラのほうにはエントリポイントのラベルが違うだ
けの⼆つの gcd列があります。これら⼆つの列を、ひとつの列 —gcdサブルー
チン (subroutine)— への分岐によって置き換えられるとしたら、そのほうがよ
いでしょう。gcdサブルーチンの末尾では、メイン命令列の正しい場所に分岐
して戻るようにします。これは、次のように実現できます。まず、gcdに分岐す
る前に、continueという特別レジスタに (0や 1といった)識別するための値
を置きます。gcd サブルーチンの末尾では、continueレジスタの値によって、
after-gcd-1または after-gcd-2に帰ります。図 5.9は、結果としてできるコ
ントローラ命令列の中で関係する部分を⽰しています。これには gcd命令列は
ひとつしかありません。

図 5.9: ↓ 図 5.8でコントローラ命令列の重複を防ぐため continue
レジスタを使う
gcd
(test (op =) (reg b) (const 0))
(branch (label gcd-done))
(assign t (op rem) (reg a) (reg b))
(assign a (reg b))
(assign b (reg t))
(goto (label gcd))

gcd-done
(test (op =) (reg continue) (const 0))
(branch (label after-gcd-1))
(goto (label after-gcd-2))
. . .

;; 最初に必要になる場所から gcdに分岐する前に
;; レジスタ continueに 0を置く
(assign continue (const 0))

(goto (label gcd))
after-gcd-1
. . .

;; ⼆回⽬に gcdを使う前にはレジスタ continueに 1を置く
(assign continue (const 1))
(goto (label gcd))

after-gcd-2

このやり⽅は、⼩さな問題を扱うにはいいですが、コントローラ命令列の中に
たくさんの gcd計算があると厄介なことになるでしょう。gcdサブルーチンの
後にどこから実⾏を続けるかを決めるためには、gcdを使う場所すべてについ
て、データパスのテストとコントローラの分岐命令を置かなければなりません。
サブルーチン実装のより強⼒な⼿法は、サブルーチンが終わった後に実⾏を続
けるべきコントローラ命令列のエントリポイントのラベルを continueレジス
タに持たせるというものです。この戦略を実装するには、レジスタマシンのデ
ータパスとコントローラの間に新しい種類の接続が必要になります。レジスタ
にコントローラ命令列のラベルを代⼊し、その値を後から取得して指定された
エントリポイントから実⾏を継続するのに使うということができるようにする
⽅法が必要です。
この機能を反映させるために、レジスタに値としてコントローラ命令列の

ラベルを (特別な定数として) 代⼊できるようにレジスタマシン⾔語の assign
命令を拡張することにします。また、goto命令も拡張して、定数ラベルで記述
したエントリポイントだけでなく、レジスタの中⾝として記述されたエントリ
ポイントからも実⾏を続けられるようにします。これらの新しい機構を使えば、
gcdサブルーチンの末尾には continueレジスタに格納された場所への分岐を
置くことができるようになります。こうすると、コントローラ列は図 5.10に⽰
すようなものになります。

図 5.10: ↓ continueレジスタにラベルを代⼊すると図 5.9の戦略を
単純化・⼀般化できる
gcd
(test (op =) (reg b) (const 0))
(branch (label gcd-done))
(assign t (op rem) (reg a) (reg b))
(assign a (reg b))

(assign b (reg t))
(goto (label gcd))

gcd-done
(goto (reg continue))
. . .

;; gcdを呼ぶ前に、gcdから帰る先のラベルを continueに代⼊す
る
(assign continue (label after-gcd-1))
(goto (label gcd))

after-gcd-1
. . .

;; ⼆つ⽬の gcd呼び出し、別の継続を持つ
(assign continue (label after-gcd-2))
(goto (label gcd))

after-gcd-2

複数のサブルーチンを持つマシンであれば、継続レジスタを複数 (gcd-
continue, factorial-continueのような)を使うことも考えられますし、すべ
てのサブルーチンでひとつの continueレジスタを共有するようにすることも
考えられます。共有したほうが経済的ですが、別のサブルーチン (sub2)を呼
び出すサブルーチン (sub1)がある場合には注意する必要があります。sub1が
sub2を呼び出すために continueを設定する前に、continueの中⾝を何かほ
かのレジスタに保存しておくということをしていなければ、sub1が終わった
時点でどこに⾏くべきかわからなくなってしまいます。再帰を扱うために次の
節で開発するメカニズムは、この⼊れ⼦のサブルーチン呼び出しという問題に
もよりよい解決法を提供してくれます。

5.1.4 再帰の実装にスタックを使う
ここまでに⽰した考え⽅を使えば、プロセスのそれぞれの状態変数に対応する
レジスタを持つレジスタマシンを規定することによって、任意の反復プロセス
を実装できます。このマシンはレジスタの中⾝を変更しながらコントローラル
ープを繰り返し実⾏し、何らかの終了条件が満たされると終了します。コント
ローラ命令列のそれぞれの位置で、(反復プロセスの状態を表現する)マシンの
状態は、レジスタの中⾝ (状態変数の値)によって完全に決定されます。
しかし、再帰プロセスを実装するにはメカニズムを追加する必要があり

ます。階乗を計算する以下の再帰的な⼿法について考えてみましょう。これ
は1.2.1節で詳しく調べたものです。
(define (factorial n)

(if (= n 1) 1 (* (factorial (- n 1)) n)))

この⼿続きを⾒るとわかるように、n!を計算するには (n− 1)!を計算する必要
があります。以下の⼿続きをモデル化した gcdマシンでも、
(define (gcd a b)

(if (= b 0) a (gcd b (remainder a b))))

やはり同じように、別の gcdを計算しなければなりませんでした。しかし、gcd
⼿続きと factorial⼿続きには重要な違いがあります。gcd⼿続きでは、元の
計算を新しい gcdの計算に簡約するのに対し、factorialでは別の階乗を部分
問題として求める必要があります。gcdの場合は、新しい gcd計算の答えが元
の問題の答えになります。次の gcdを計算するには、単純に新しい引数を gcd
マシンの⼊⼒レジスタに置いて同じコントローラ命令列を実⾏することによっ
てマシンのデータパスを再利⽤します。マシンが最後の gcd問題を解き終わる
と、計算全体が終わったことになります。
階乗 (に限らず、任意の再帰プロセス)では、新しい階乗の部分問題の答え

は元の問題の答えではありません。(n− 1)!について得られた値は、nをかけて
最終的な答えを得る必要があります。gcdの設計の真似をして、レジスタ nを
デクリメントして階乗マシンを再び動かすことで部分問題を解こうとしても、
結果にかけるべき古い nの値はもうありません。ですので、部分問題を解くに
は⼆つ⽬の階乗マシンが必要です。この⼆つ⽬の階乗計算はそれ⾃⾝階乗の部
分問題を持っていて、それには三つ⽬の階乗マシンが必要です。ずっとこの繰
り返しです。それぞれの階乗マシンは中に別の階乗マシンを持っているので、
マシン全体は同じようなマシンの無限の⼊れ⼦を持つことになり、有限の決ま
った数の部品からは構築できないということになります。
しかしながら、⼊れ⼦になったそれぞれのマシンに対して同じ部品を使う

ようにできれば、階乗プロセスをレジスタマシンとして実装できます。具体的
には、n!を計算するマシンは (n− 1)!を計算する部分問題、(n− 2)!の部分問
題 . . .にも同じ部品を使うようにするということです。階乗プロセスは計算を
⾏うために限りなくたくさんの同じマシンのコピーが必要だということになっ
ていますが、ある特定の時間にアクティブである必要があるのはこれらのコピ
ーのうちのひとつだけなので、これは実現できそうです。マシンが再帰部分問
題にぶつかると、主問題の作業を中断し、同じ物理部品を使い回して部分問題
を処理し、それから中断した計算を続けるというようにできます。

after-
fact

fact-
done

(controller
  (assign continue (label fact-done))   ;set up final return address
fact-loop
  (test (op =) (reg n) (const 1))
  (branch (label base-case))
  ;; Set up for the recursive call by saving n and continue.
  ;; Set up continue so that the computation will continue
  ;; at after-fact when the subroutine returns.
  (save continue)
  (save n)
  (assign n (op -) (reg n) (const 1))
  (assign continue (label after-fact))
  (goto (label fact-loop))
after-fact
  (restore n)
  (restore continue)
  (assign val (op *) (reg n) (reg val))  ;val now contains n(n - 1)!
  (goto (reg continue))                  ;return to caller
base-case
  (assign val (const 1))                 ;base case: 1! = 1
  (goto (reg continue))                  ;return to caller
fact-done)

1

=

val n stack

continue* --

sn

rn
scrc

controller

図 5.11: 再帰階乗マシン

部分問題の中では、レジスタの中⾝が主問題のときと異なるものになりま
す (ここでは、レジスタ nはデクリメントされます)。中断した計算を続けられ
るようにするには、マシンは部分問題を解いた後に必要になるレジスタの中⾝
を保存して、それらを復元して中断した計算を続けられるようにする必要があ
ります。階乗の場合、デクリメントしたレジスタ nの階乗の計算が終わったと
きに復元できるように、nの古い値を保存しておくことになります。2

⼊れ⼦の再帰呼び出しの深さにあらかじめ決められた限界はないため、任
意の数のレジスタの値を保存することが必要になる可能性があります。それら
の値は、保存したのとは逆順に復元しなければなりません。⼊れ⼦の再帰では、
最後に⼊った部分問題が最初に終わるからです。このため、レジスタの値の保
存には、スタック (stack)、つまり「後⼊れ先出し」のデータ構造を使うことに
なります。レジスタマシン⾔語にスタックを含めることは、⼆つの命令を追加
することによって⾏えます。ひとつは save命令で、これはスタックに値を追
加するというものです。もうひとつは restore命令で、これによってスタック
から値を復元します。⼀連の値をスタックに saveした後に、restore命令を
連続して実⾏すると、値が逆順に取り出されます。3

スタックの助けを借りると、ひとつの階乗マシンのデータパスをそれぞれ
の階乗部分問題に使い回せるようになります。データパスを操作するコントロ
ーラ命令列を再利⽤する際にも、同じような設計上の問題があります。階乗計
算を再実⾏するには、コントローラは反復プロセスのように単純に最初の場所
までループで戻るというわけにはいきません。(n − 1)!という部分問題を解い
た後に、マシンはその結果に nをかける必要があるからです。コントローラは
n!の計算を中断し、部分問題 (n − 1)!を解き、それから n!の計算を続けなけ
ればなりません。このように階乗演算を⾒ると、5.1.3節で述べたサブルーチ
ン機構を使うのがよさそうです。サブルーチン機構とは、部分問題を解く部分
に移動した後で主問題の残りの部分を続けるために、コントローラがレジスタ
continueを使うようにするというものでした。この考え⽅で、continueレジ
スタに格納されたエントリポイントに戻る階乗サブルーチンを作ることができ
ます。それぞれのサブルーチン呼び出しの前後では、nレジスタと同じように
continueの保存・復元を⾏うことにします。これは、階乗計算のそれぞれの

2ここで古い nを保存する必要はないのではないかと思われるかもしれません。nをデ
クリメントして部分問題を解いたら、単純にそれをインクリメントして元の値を復元す
ることもできます。しかし、この戦略は階乗ではうまくいくものの、⼀般的にはうまく
いきません。レジスタの古い値が常に新しい値から計算できるとは限らないからです。

35.3節では、より基本的な演算を使ってスタックを実装する⽅法について⾒ていきま
す。

555



“レベル”で、同じ continueレジスタを使うことになるからです。つまり、階
乗サブルーチンは部分問題を解くために⾃分⾃⾝を呼び出すときには新しい値
を continueに⼊れる必要がありますが、⾃分の担当する部分問題を解いた後
に呼び出した場所に戻るには元の値が必要だということです。
図 5.11に、再帰 factorial⼿続きを実装するマシンのデータパスとコント

ローラを⽰します。このマシンは、ひとつのスタックと、n, val, continueと
いう三つのレジスタを持っています。データパス図を単純化するために、レジ
スタ代⼊ボタンには名前をつけずに、スタック演算ボタン (レジスタを保存す
る scと sn、レジスタを復元する rcと rn)だけ名前をつけています。マシンを
操作するには、階乗を求めたい数をレジスタ nに⼊れ、マシンをスタートさせ
ます。マシンが fact-doneに着くと計算は完了で、答えはレジスタ valに⼊っ
ています。コントローラ命令列では、nと continueがそれぞれの再帰呼び出し
の前に保存され、その呼び出しから戻るときに復元されます。呼び出しから戻
るのは、continueに格納された場所に分岐することによって実現されていま
す。マシンがスタートするときに、最後に帰る場所が fact-doneになるように
continueが初期化されます。階乗計算の結果を持つ valレジスタは再帰呼び
出しの前に保存されません。サブルーチンから帰った後には、valの古い中⾝
は役に⽴たないためです。必要なのは、部分問題によって⽣成された新しい値
だけです。
階乗計算には原理上は無限のマシンが必要ですが、実際には図 5.11のマシ

ンは、無限に⼤きくなりうるスタック以外は有限です。しかし、スタックを物
理的に実装したものはどうやっても有限サイズのものになるため、このマシン
で扱える再帰呼び出しの深さはそれによって縛られることになります。この階
乗の実装は、スタックという拡張を加えた通常のレジスタマシンによって再帰
アルゴリズムを実現する⼀般的な戦略を⽰しています。再帰部分問題にぶつか
ると、部分問題を解いた後に現在の値が必要となるレジスタをスタックに保存
し、再帰部分問題を解き、保存したレジスタを復元して主問題の実⾏を続けま
す。continueレジスタは常に保存する必要があります。保存するべきレジスタ
がほかにあるかどうかはマシンによります。再帰計算のすべてが、部分問題を
解く間に変更されるレジスタの元の値を必要とするわけではないからです (練
習問題 5.4参照)。

⼆重再帰
より複雑な再帰プロセスである、1.2.2節で紹介したフィボナッチ数の⽊の再
帰の計算について考えてみましょう。

(define (fib n)
(if (< n 2)

n
(+ (fib (- n 1)) (fib (- n 2)))))

階乗と同じように、再帰フィボナッチ計算はレジスタ n, val, continueを持つ
レジスタマシンとして実装できます。このマシンは階乗マシンよりも複雑です。
コントローラ命令列の中で再帰呼び出しの実⾏が必要となる場所が⼆箇所ある
からです。ひとつは Fib(n− 1)を求めるところで、もうひとつは Fib(n− 2)を
求めるところです。それぞれの呼び出しの準備として、後で値が必要になるレ
ジスタを保存し、再帰的にフィボナッチ数を求める数 (n − 1または n − 2)を
nレジスタに設定し、continueに戻り先の主命令列のエントリポイント (それ
ぞれ afterfib-n-1または afterfib-n-2)を代⼊します。それから fib-loop
に⾏きます。再帰呼び出しから帰るときには、答えは valに⼊っています。図
5.12にこのマシンのコントローラ命令列を⽰します。

図 5.12: ↓ フィボナッチ数を計算するマシンのコントローラ
(controller

(assign continue (label fib-done))
fib-loop

(test (op <) (reg n) (const 2))
(branch (label immediate-answer))
;; Fib(n− 1)を求める準備
(save continue)
(assign continue (label afterfib-n-1))
(save n) ; nの古い値を保存
(assign n (op -) (reg n) (const 1)) ; n を n-1 で

上書き
(goto (label fib-loop)) ; 再帰呼び出しの実⾏

afterfib-n-1 ; リターン時に Fib(n− 1)は valに⼊って
いる

(restore n)
(restore continue)
;; Fib(n− 2)を求める準備
(assign n (op -) (reg n) (const 2))
(save continue)
(assign continue (label afterfib-n-2))

(save val) ; Fib(n− 1)を保存
(goto (label fib-loop))

afterfib-n-2 ; リターン時に Fib(n− 2)は valに⼊って
いる

(assign n (reg val)) ; nには Fib(n− 2)が⼊る
(restore val) ; valには Fib(n− 1)が⼊る
(restore continue)
(assign val ; Fib(n− 1) + Fib(n− 2)

(op +) (reg val) (reg n))
(goto (reg continue)) ; 呼び出し元に戻る、答えは

valの中
immediate-answer

(assign val (reg n)) ; 基底の場合: Fib(n) = n

(goto (reg continue))
fib-done)

練習問題 5.4: 次のそれぞれの⼿続きを実装するレジスタマシンを
規定せよ。それぞれのマシンについて、コントローラ命令列を書
き、データパスを⽰す図を描け。

a. 再帰的指数計算
(define (expt b n)

(if (= n 0)
1
(* b (expt b (- n 1)))))

b. 反復的指数計算
(define (expt b n)

(define (expt-iter counter product)
(if (= counter 0)

product
(expt-iter (- counter 1)

(* b product))))
(expt-iter n 1))

練習問題 5.5: 何らかの⾃明でない (少なくとも⼀回は再帰呼び出
しの実⾏が必要となるような)⼊⼒を使って、階乗マシンとフィボ
ナッチマシンを⼿作業でシミュレートせよ。実⾏中の重要な場所
でのスタックの中⾝を⽰せ。
練習問題 5.6: Ben Bitdiddleは、フィボナッチマシンのコントロー
ラ命令列に余分な saveと restoreがあり、取り除くことによって
マシンを⾼速化できることに気がついた。それらの命令はどれか。

5.1.5 命令まとめ
レジスタマシン⾔語のコントローラ命令は、以下のいずれかの形式をとります。
⟨inputi⟩はそれぞれ (reg<register-name>)か (const <constant-value>)の
いずれかです。これらの命令は5.1.1節で導⼊したものです。
(assign ⟨register-name⟩ (reg ⟨register-name⟩))
(assign ⟨register-name⟩ (const ⟨constant-value⟩))
(assign ⟨register-name⟩

(op ⟨operation-name⟩)
⟨input1⟩ . . . ⟨inputn⟩)

(perform (op ⟨operation-name⟩) ⟨input1⟩ . . . ⟨inputn⟩)
(test (op ⟨operation-name⟩) ⟨input1⟩ . . . ⟨inputn⟩)
(branch (label ⟨label-name⟩))
(goto (label ⟨label-name⟩))

ラベルを保存するのにレジスタを使うことについては5.1.3節で導⼊しました。
(assign ⟨register-name⟩ (label ⟨label-name⟩))
(goto (reg ⟨register-name⟩))

スタックを使う命令は5.1.4節で導⼊しました。
(save ⟨register-name⟩)
(restore ⟨register-name⟩)

ここまでで⾒た ⟨constant-value⟩ の種類は数値だけですが、これから先では、
⽂字列、記号、リストも使うことになります。
(const "abc") これは⽂字列 "abc",
(const abc) これは記号 abc,

(const (a b c)) これはリスト (a b c),
(const ()) これは空リスト

5.2 レジスタマシンシミュレータ
レジスタマシンの設計をよく理解するには、設計したマシンが期待通りに動く
かどうかを⾒るためにテストを⾏う必要があります。設計をテストする⽅法の
ひとつとして、練習問題 5.5でやったように、コントローラの演算を⼿作業で
シミュレートするというものがあります。しかしこれは、よほど簡単なマシン
でもなければ、⾮常に退屈な作業になります。この章では、レジスタマシン⾔
語で記述されたマシンのシミュレータを構築します。シミュレータは四つのイ
ンターフェイス⼿続きを持つ Schemeのプログラムです。⼀つ⽬のインターフ
ェイス⼿続きは、レジスタマシンの記述を使ってマシンのモデル (それぞれの
部品がシミュレート対象のマシンの部品に対応するデータ構造)を構築します。
残り三つは、モデルを操作してマシンをシミュレートできるようにするための
ものです。

(make-machine ⟨register-names⟩ ⟨operations⟩ ⟨controller⟩)

この⼿続きは、与えられたレジスタ、演算、コントローラを持つマ
シンのモデルを構築し、それを返します。
(set-register-contents! ⟨machine-model⟩

⟨register-name⟩
⟨value⟩)

この⼿続きは、与えられたマシンの (シミュレートした)レジスタ
に値を格納します。
(get-register-contents ⟨machine-model⟩ ⟨register-name⟩)

この⼿続きは、与えられたマシンの (シミュレートした)レジスタ
の中⾝を返します。
(start ⟨machine-model⟩)

この⼿続きは、与えられたマシンの実⾏をシミュレートします。実
⾏はコントローラ命令列の最初からスタートし、命令列の末尾に
着くと⽌まります。

これらの⼿続きの使い⽅の例として、5.1.1 節の gcd マシンのモデルとして
gcd-machineを以下のように定義します。
(define gcd-machine

(make-machine
'(a b t)
(list (list 'rem remainder) (list '= =))
'(test-b (test (op =) (reg b) (const 0))

(branch (label gcd-done))
(assign t (op rem) (reg a) (reg b))
(assign a (reg b))
(assign b (reg t))
(goto (label test-b))
gcd-done)))

make-machineの⼀つ⽬の引数はレジスタ名のリストです。次の引数は、演算
名とその演算を実装した (つまり、同じ⼊⼒値を与えられると同じ出⼒値を⽣
成する)Scheme⼿続きをペアにしたテーブル (2要素リストのリスト)です。最
後の引数は、ラベルとマシン命令のリストとしてコントローラを規定します。
これは5.1節で⽰したものと同じです。
このマシンを使って gcdを計算するには、⼊⼒レジスタを設定し、マシン

をスタートさせ、シミュレーションが終了したら結果を調べます。
(set-register-contents! gcd-machine 'a 206)
done
(set-register-contents! gcd-machine 'b 40)
done
(start gcd-machine)
done
(get-register-contents gcd-machine 'a)
2

この計算は、Schemeで書いた gcd⼿続きよりもずっと遅く動作します。これ
は、assignのような低レベルのマシン命令をずっと複雑な演算によってシミ
ュレートすることになるからです。

練習問題 5.7: シミュレータを使って練習問題 5.4で設計したマシ
ンをテストせよ。

5.2.1 マシンモデル
make-machineで⽣成するマシンのモデルは、第 3章で開発したメッセージパ
ッシング技術を使って、局所状態を持つ⼿続きとして表現されます。このモデ
ルを構築するために、make-machineはまず make-new-machine ⼿続きを呼ん
で、すべてのレジスタマシンに共通なマシンモデルの部品を構築します。make-
new-machineによって構築されるこの基本マシンモデルは、本質的には複数の
レジスタとスタックひとつに、コントローラ命令をひとつずつ処理する実⾏メ
カニズムをセットにする⼊れ物です。
次に、make-machineは (メッセージを送って)この基本モデルを拡張して、

レジスタ、演算、定義中の特定のマシンのコントローラを持たせます。最初に、
与えられたレジスタ名それぞれについて新しいマシンの中にレジスタを割り当
て、指定された演算をマシンに組み込みます。次に、アセンブラ (assembler)(後
ほど 5.2.2節で説明します)を使ってコントローラ命令列を新しいマシンの命
令に変換し、その命令列をマシンの命令列として組み込みます。make-machine
は、修正したマシンのモデルを値として返します。
(define (make-machine register-names ops controller-text)

(let ((machine (make-new-machine)))
(for-each
(lambda (register-name)

((machine 'allocate-register) register-name))
register-names)

((machine 'install-operations) ops)
((machine 'install-instruction-sequence)
(assemble controller-text machine))

machine))

レジスタ
レジスタは、第 3章でやったように、局所状態を持つ⼿続きとして表現します。
⼿続き make-registerはアクセスと変更が可能な値を持つレジスタを作成し
ます。

(define (make-register name)
(let ((contents '*unassigned*))

(define (dispatch message)
(cond ((eq? message 'get) contents)

((eq? message 'set)
(lambda (value) (set! contents value)))

(else
(error "Unknown request: REGISTER" message))))

dispatch))

以下の⼆つの⼿続きがレジスタにアクセスするために使われます。
(define (get-contents register) (register 'get))
(define (set-contents! register value)

((register 'set) value))

スタック
スタックも局所状態を持つ⼿続きとして表現できます。⼿続き make-stackは
スタックを作ります。スタックの局所状態はスタック上の項⽬のリストです。
スタックは、スタックにアイテムを追加する push、スタックのトップから項⽬
を取り去りそれを返す pop、スタックを空に初期化する initializeというリ
クエストを受け付けます。
(define (make-stack)

(let ((s '()))
(define (push x) (set! s (cons x s)))
(define (pop)

(if (null? s)
(error "Empty stack: POP")
(let ((top (car s)))

(set! s (cdr s))
top)))

(define (initialize)
(set! s '())
'done)

(define (dispatch message)
(cond ((eq? message 'push) push)

((eq? message 'pop) (pop))
((eq? message 'initialize) (initialize))
(else (error "Unknown request: STACK" message))))

dispatch))

以下の⼆つの⼿続きがスタックへのアクセスに使われます。
(define (pop stack) (stack 'pop))
(define (push stack value) ((stack 'push) value))

基本マシン
図 5.13に⽰す make-new-machine⼿続きは、ひとつのスタック、初期値が空の
命令列、初期値としてスタックを空にする演算を持つ演算のリスト、初期値と
して flagと pc(「プログラムカウンタ」)という名前の⼆つのレジスタを持つ
レジスタテーブル (register table)を局所状態として持つオブジェクトを構築し
ます。内部⼿続き allocate-registerはレジスタテーブルに新しい項⽬を追
加し、内部⼿続き lookup-registerはテーブルからレジスタを検索します。

flagレジスタは、シミュレート対象のマシンの分岐を制御するために使い
ます。test命令はテストの結果 (真または偽)を flagの中⾝として設定しま
す。branch命令は flagの中⾝を調べて分岐するかどうかを決めます。

pcレジスタはマシン動作中の命令の順序を決定します。命令の逐次実⾏は
内部⼿続き executeとして実装されています。このシミュレーションモデルで
は、それぞれのマシン命令は引数なしの⼿続きを持つデータ構造で、命令実⾏
⼿続き (instruction execution procedure)と呼ばれます。命令実⾏⼿続きを呼ぶ
と命令の実⾏をシミュレートします。シミュレーションの実⾏中、pcは次に実
⾏される命令から始まる命令列の場所を指します。executeはその命令を取得
し、命令実⾏⼿続きを呼んでそれを実⾏し、実⾏する命令がなくなるまで (つ
まり、pcが命令列の末尾を指すまで)このサイクルを繰り返します。

図 5.13: ↓ 基本マシンモデルを実装する make-new-machine⼿続き
(define (make-new-machine)

(let ((pc (make-register 'pc))
(flag (make-register 'flag))
(stack (make-stack))
(the-instruction-sequence '()))

(let ((the-ops
(list (list 'initialize-stack

(lambda () (stack 'initialize)))))
(register-table
(list (list 'pc pc) (list 'flag flag))))

(define (allocate-register name)
(if (assoc name register-table)

(error "Multiply defined register: " name)
(set! register-table

(cons (list name (make-register name))
register-table)))

'register-allocated)
(define (lookup-register name)

(let ((val (assoc name register-table)))
(if val

(cadr val)
(error "Unknown register:" name))))

(define (execute)
(let ((insts (get-contents pc)))

(if (null? insts)
'done
(begin

((instruction-execution-proc (car insts)))
(execute)))))

(define (dispatch message)
(cond ((eq? message 'start)

(set-contents! pc the-instruction-sequence)
(execute))

((eq? message 'install-instruction-sequence)
(lambda (seq)

(set! the-instruction-sequence seq)))
((eq? message 'allocate-register)
allocate-register)

((eq? message 'get-register)
lookup-register)

((eq? message 'install-operations)
(lambda (ops)

(set! the-ops (append the-ops ops))))
((eq? message 'stack) stack)
((eq? message 'operations) the-ops)
(else (error "Unknown request: MACHINE"

message))))
dispatch)))

それぞれの命令実⾏⼿続きは、演算の⼀環として、pcを変更して次に実⾏す
るべき命令を指すようにします。branchと goto命令は新しい⾏き先を指すよ
うに pcを変更します。これら以外の命令はすべて、単純に pcを進めて命令
列中の次の命令を指すようにします。それぞれの executeの呼び出しが再び
executeを呼び出すことに注意してください。これは無限ループにはなりませ
ん。命令実⾏⼿続きの実⾏は pcの中⾝を変更するからです。

make-new-machineはメッセージパッシングによるマシンの内部状態への
アクセスを実装した dispatch⼿続きを返します。マシンのスタートは、命令
列の先頭を pcに設定し、executeを呼ぶことによって⾏えます。
マシンを扱いやすくするために、マシンの start演算の代替として⼿続き

によるインターフェイスを提供します。ほかに、レジスタの中⾝の設定と検査
の⼿続きも、5.2節の最初で規定したような形で提供します。
(define (start machine) (machine 'start))
(define (get-register-contents machine register-name)

(get-contents (get-register machine register-name)))
(define (set-register-contents! machine register-name value)

(set-contents! (get-register machine register-name)
value)

'done)

これらの⼿続き (また5.2.2節と5.2.3節の多くの⼿続き)は、あるマシンのある
レジスタ名のレジスタを探すために次の⼿続きを使います。

(define (get-register machine reg-name)
((machine 'get-register) reg-name))

5.2.2 アセンブラ
アセンブラは、あるマシンのコントローラの式の列をそれに対応するマシン命
令の列に変換するものです。マシン命令はそれぞれ実⾏⼿続きを持ちます。全
体として、アセンブラは 第 4章で詳しく⾒てきた評価器によく似ています。ど
ちらも⼊⼒⾔語 (この場合はレジスタマシン⾔語)があり、その⾔語の式のそ
れぞれの型に対して適切なアクションを実⾏する必要があります。
それぞれの命令に対する実⾏⼿続きを⽣成する技術は、4.1.7節で実⾏時の

処理から解析を分離して評価器の速度向上をするのに使ったものと同じです。
第 4章で⾒てきたように、Scheme式の有⽤な解析の多くは、変数の実際の値
を知らないでも⾏うことができます。同じように、レジスタマシン⾔語の式も、
マシンレジスタの実際の中⾝を知らないままで多くの有⽤な解析が⾏えます。
例えば、レジスタへの参照はレジスタオブジェクトへのポインタに置き換える
ことができますし、ラベルへの参照はそのラベルが指⽰する命令列の場所への
ポインタに置き換えることができます。
アセンブラは、命令実⾏⼿続きを⽣成する前に、ラベルの参照先をすべて

知っておく必要があります。そのため、アセンブラはまずコントローラのテキ
ストを⾛査して、命令からラベルを分離します。アセンブラがテキストを⾛査
する間に、命令リストと、各ラベルとそのリストへのポインタを関連づけるテ
ーブルの両⽅を構築します。それから、アセンブラは命令リストのそれぞれの
命令に実⾏⼿続きを挿⼊することで命令リストを拡張します。

assemble ⼿続きはアセンブラのメインエントリです。コントローラテキ
ストとマシンモデルを引数に取り、モデルに格納するべき命令列を返します。
assembleは extract-labelsを呼び、与えられたコントローラテキストから最
初の命令リストとラベルテーブルを構築します。extract-labelsの⼆つ⽬の
引数は、これらの結果を処理するために呼ぶ⼿続きです。この⼿続きは update-
insts!を使って命令実⾏⼿続きを⽣成し、それらを命令リストに挿⼊し、修正
したリストを返します。
(define (assemble controller-text machine)

(extract-labels
controller-text
(lambda (insts labels)

(update-insts! insts labels machine)
insts)))

extract-labelsは、引数として textというリスト (コントローラ命令式の列)
と、receive⼿続きを取ります。receiveは、次の⼆つの値を引数として呼び
出されます。(1) 命令データ構造のリスト insts。それぞれの命令データ構造
は text内の命令を含みます。(2) テーブル labels。textの各ラベルと、その
ラベルが指⽰するリスト insts内の位置を関連づけます。
(define (extract-labels text receive)

(if (null? text)
(receive '() '())
(extract-labels
(cdr text)
(lambda (insts labels)

(let ((next-inst (car text)))
(if (symbol? next-inst)

(receive insts
(cons (make-label-entry next-inst

insts)
labels))

(receive (cons (make-instruction next-inst)
insts)

labels)))))))

extract-labelsは、textの要素を順に⾛査し、instsと labelsを集積してい
きます。要素が記号 (つまりラベル)であれば、labelsテーブルに適切なエン
トリが追加されます。そうでなければ、要素は instsリストに集積されます。4

4ここで receive⼿続きを使っているのは、extract-labelsに labelsと instsとい
う⼆つの値を、それらを保持する複合データ構造を明⽰的に作らずに効率的に返させる
ためです。明⽰的に値のペアを返すようにした実装は以下のようになります。
(define (extract-labels text)

(if (null? text)
(cons '() '())
(let ((result (extract-labels (cdr text))))

(let ((insts (car result)) (labels (cdr result)))
(let ((next-inst (car text)))

(if (symbol? next-inst)

update-insts!は命令リストを修正します。命令リストは初期状態では命
令のテキストしか持っていませんが、それに対応する実⾏⼿続きを追加します。
(define (update-insts! insts labels machine)

(let ((pc (get-register machine 'pc))
(flag (get-register machine 'flag))
(stack (machine 'stack))
(ops (machine 'operations)))

(for-each
(lambda (inst)

(set-instruction-execution-proc!
inst
(make-execution-procedure
(instruction-text inst)
labels machine pc flag stack ops)))

insts)))

マシン命令データ構造は、命令テキストと、それに対応する実⾏⼿続きを単純
にペアにします。実⾏⼿続きは extract-labelsが命令を構築するときにはま
だ使⽤できず、後で update-insts!によって挿⼊されます。
(define (make-instruction text) (cons text '()))

(cons insts
(cons (make-label-entry next-inst insts)

labels))
(cons (cons (make-instruction next-inst) insts)

labels)))))))

これは, assembleからは次のように呼び出すことになります。
(define (assemble controller-text machine)

(let ((result (extract-labels controller-text)))
(let ((insts (car result)) (labels (cdr result)))

(update-insts! insts labels machine)
insts)))

ここでの receiveの使い⽅は、複数の値を返すエレガントな⽅法の例と考えていただい
てもいいですし、プログラミングトリックをひけらかしたいだけだと考えていただいて
もけっこうです。receiveのような、次に呼び出すべき⼿続きを渡す引数は、“継続”と
呼ばれます。継続は4.3.3節で amb評価器のバックトラック制御構造の実装にも使ったこ
とを思い出しましょう。

(define (instruction-text inst) (car inst))
(define (instruction-execution-proc inst) (cdr inst))
(define (set-instruction-execution-proc! inst proc)

(set-cdr! inst proc))

命令テキストはシミュレータには使われませんが、デバッグ⽤に持っておくと
便利です (練習問題 5.16参照)。
ラベルテーブルの要素はペアです。

(define (make-label-entry label-name insts)
(cons label-name insts))

テーブル内の要素は以下にように検索します。
(define (lookup-label labels label-name)

(let ((val (assoc label-name labels)))
(if val

(cdr val)
(error "Undefined label: ASSEMBLE"

label-name))))

練習問題 5.8: 以下のレジスタマシンのコードは、ラベル hereが
複数回定義されているため曖昧になっている。
start

(goto (label here))
here

(assign a (const 3))
(goto (label there))

here
(assign a (const 4))
(goto (label there))

there

本⽂に書いたようなシミュレータでは、制御が there に達した
とき、レジスタ a の中⾝はどうなっているか。⼿続き extract-
labelsを修正し、同じラベル名が⼆つの異なる場所を指すために
使われている場合にはエラーを発するようにせよ。

570



5.2.3 命令の実⾏⼿続きの⽣成
アセンブラは make-execution-procedureを呼んで命令の実⾏⼿続きを⽣成し
ます。4.1.7節の評価器の analyze⼿続きと同じように、この⼿続きは適切な
実⾏⼿続きを⽣成するために命令の型によってディスパッチを⾏います。
(define (make-execution-procedure

inst labels machine pc flag stack ops)
(cond ((eq? (car inst) 'assign)

(make-assign inst machine labels ops pc))
((eq? (car inst) 'test)
(make-test inst machine labels ops flag pc))

((eq? (car inst) 'branch)
(make-branch inst machine labels flag pc))

((eq? (car inst) 'goto)
(make-goto inst machine labels pc))

((eq? (car inst) 'save)
(make-save inst machine stack pc))

((eq? (car inst) 'restore)
(make-restore inst machine stack pc))

((eq? (car inst) 'perform)
(make-perform inst machine labels ops pc))

(else
(error "Unknown instruction type: ASSEMBLE"

inst))))

レジスタマシン⾔語の命令のそれぞれの型に対して、適切な実⾏⼿続きを構築
する⽣成器があります。これらの⼿続きの詳細が、レジスタマシン⾔語のそれ
ぞれの命令の構⽂と意味の両⽅を決定します。ここでは、データ抽象化を使っ
てレジスタマシンの式の詳細な構⽂を全体的な実⾏メカニズムから分離してい
ます。これは、4.1.2節の評価器で、構⽂⼿続きを使って命令のそれぞれの部分
を抽出・分類したのと同じです。

assign命令
make-assign⼿続きは assign命令を処理します。
(define (make-assign inst machine labels operations pc)

(let ((target
(get-register machine (assign-reg-name inst)))

(value-exp (assign-value-exp inst)))
(let ((value-proc

(if (operation-exp? value-exp)
(make-operation-exp
value-exp machine labels operations)

(make-primitive-exp
(car value-exp) machine labels))))

(lambda () ; assignに対する実⾏⼿続き
(set-contents! target (value-proc))
(advance-pc pc)))))

make-assignは、assign命令から代⼊先レジスタ名 (命令の⼆つ⽬の要素)と
値の式 (命令を構成するリストの残りの部分)を次の⼆つのセレクタを使って
抽出します。
(define (assign-reg-name assign-instruction)

(cadr assign-instruction))
(define (assign-value-exp assign-instruction)

(cddr assign-instruction))

レジスタ名は get-registerによって検索され、代⼊先となるレジスタオブジ
ェクトを⽣成します。値の式は、値が命令の結果であれば make-operation-exp
に渡され、そうでなければ make-primitive-expに渡されます。これらの⼿続
き (以下に⽰します)は、値の式を構⽂解析し、その値に対する実⾏⼿続きを⽣
成します。これは value-procと呼ばれる引数なしの⼿続きで、シミュレーシ
ョンの間に評価されてレジスタに代⼊すべき実際の値を⽣成します。レジスタ
名の検索と値の式の構⽂解析はアセンブリ時に⼀回⾏われるだけで、命令をシ
ミュレーションするたびに毎回⾏われるわけではないというところに注意して
ください。この省⼒化こそが実⾏⼿続きを使う理由です。これは、4.1.7節の評
価器で、プログラム分析を実⾏から分離することによって達成した省⼒化に直
接対応します。

make-assignによって返される結果は assign命令の実⾏⼿続きです。この
⼿続きが (マシンモデルの execute⼿続きによって)呼ばれると、value-proc
⼿続きを実⾏することで得られる結果を代⼊先レジスタの中⾝に設定します。
それから、以下の⼿続きを実⾏して pcを次の命令に進めます。

(define (advance-pc pc)
(set-contents! pc (cdr (get-contents pc))))

advance-pcは、branchと goto以外のすべての命令の正常終了処理です。

test, branch, goto命令
make-testは、make-assignと同じように test命令を処理します。make-test
はテスト対象の条件を規定する式を抽出し、それに対する実⾏⼿続きを⽣成し
ます。シミュレーション時には、条件の⼿続きが呼ばれ、結果が flagレジス
タに割り当てられ、pcが進められます。

(define (make-test inst machine labels operations flag pc)
(let ((condition (test-condition inst)))

(if (operation-exp? condition)
(let ((condition-proc

(make-operation-exp
condition machine labels operations)))

(lambda ()
(set-contents! flag (condition-proc))
(advance-pc pc)))

(error "Bad TEST instruction: ASSEMBLE" inst))))
(define (test-condition test-instruction)

(cdr test-instruction))

branch命令の実⾏⼿続きは、flagレジスタの中⾝をチェックし、分岐の⽬的
地を pcの中⾝に設定する (分岐する場合)か、単に pcを進める (分岐しない場
合)かのどちらかを⾏います。branch命令内で指定される⽬的値はラベルでな
ければならず、make-branch⼿続きはそれをチェックしています。また、ラベ
ルはアセンブリ時に検索され、branch命令がシミュレートされるたびに毎回
検索されるわけではないというところにも注意してください。
(define (make-branch inst machine labels flag pc)

(let ((dest (branch-dest inst)))
(if (label-exp? dest)

(let ((insts
(lookup-label

labels
(label-exp-label dest))))

(lambda ()
(if (get-contents flag)

(set-contents! pc insts)
(advance-pc pc))))

(error "Bad BRANCH instruction: ASSEMBLE" inst))))
(define (branch-dest branch-instruction)

(cadr branch-instruction))

goto命令は branchに似ていますが、⽬的地がラベルまたはレジスタによって
規定されるという点と、チェックするべき条件がないというところが違います。
pcは常に新しい⽬的地に設定されます。
(define (make-goto inst machine labels pc)

(let ((dest (goto-dest inst)))
(cond ((label-exp? dest)

(let ((insts (lookup-label
labels
(label-exp-label dest))))

(lambda () (set-contents! pc insts))))
((register-exp? dest)
(let ((reg (get-register

machine
(register-exp-reg dest))))

(lambda ()
(set-contents! pc (get-contents reg)))))

(else (error "Bad GOTO instruction: ASSEMBLE" inst)))))
(define (goto-dest goto-instruction)

(cadr goto-instruction))

その他の命令
スタック命令の save と restore は、指定されたレジスタとスタックを使い、
pcを進めるだけです。
(define (make-save inst machine stack pc)

(let ((reg (get-register machine

(stack-inst-reg-name inst))))
(lambda ()

(push stack (get-contents reg))
(advance-pc pc))))

(define (make-restore inst machine stack pc)
(let ((reg (get-register machine

(stack-inst-reg-name inst))))
(lambda ()

(set-contents! reg (pop stack))
(advance-pc pc))))

(define (stack-inst-reg-name stack-instruction)
(cadr stack-instruction))

make-performが扱う最後の命令型は、実⾏するべきアクションに対する実⾏
⼿続きを⽣成します。シミュレーション時には、このアクション⼿続きが実⾏
され、pcが進められます。
(define (make-perform inst machine labels operations pc)

(let ((action (perform-action inst)))
(if (operation-exp? action)

(let ((action-proc
(make-operation-exp
action machine labels operations)))

(lambda () (action-proc) (advance-pc pc)))
(error "Bad PERFORM instruction: ASSEMBLE" inst))))

(define (perform-action inst) (cdr inst))

部分式の実⾏⼿続き
reg, label, const式の値は、レジスタへの代⼊ (make-assign)または演算に対
する⼊⼒ (下記の make-operation-exp)のために必要になることがあります。
以下の⼿続きは、これらの式の値をシミュレーション時に⽣成する実⾏⼿続き
を⽣成します。
(define (make-primitive-exp exp machine labels)

(cond ((constant-exp? exp)
(let ((c (constant-exp-value exp)))

(lambda () c)))

((label-exp? exp)
(let ((insts (lookup-label

labels
(label-exp-label exp))))

(lambda () insts)))
((register-exp? exp)
(let ((r (get-register machine (register-exp-reg exp))))

(lambda () (get-contents r))))
(else (error "Unknown expression type: ASSEMBLE" exp))))

reg, label, const式の構⽂は以下の⼿続きによって決められています。
(define (register-exp? exp) (tagged-list? exp 'reg))
(define (register-exp-reg exp) (cadr exp))
(define (constant-exp? exp) (tagged-list? exp 'const))
(define (constant-exp-value exp) (cadr exp))
(define (label-exp? exp) (tagged-list? exp 'label))
(define (label-exp-label exp) (cadr exp))

assign, perform, test 命令は、(op 式の形の) マシンの演算命令を、(reg や
const 式の形で与えられる) 被演算⼦に適⽤しなければならない場合があり
ます。以下の⼿続きは、“演算式”—命令内の演算と被演算⼦式からなるリス
ト—に対する実⾏⼿続きを⽣成します。
(define (make-operation-exp exp machine labels operations)

(let ((op (lookup-prim (operation-exp-op exp)
operations))

(aprocs
(map (lambda (e)

(make-primitive-exp e machine labels))
(operation-exp-operands exp))))

(lambda ()
(apply op (map (lambda (p) (p)) aprocs)))))

演算式の構⽂は以下の⼿続きによって決められています。
(define (operation-exp? exp)

(and (pair? exp) (tagged-list? (car exp) 'op)))
(define (operation-exp-op operation-exp)

(cadr (car operation-exp)))
(define (operation-exp-operands operation-exp)

(cdr operation-exp))

演算式の処理は、4.1.7節の評価器の analyze-application⼿続きによる⼿続
き適⽤の処理でそれぞれの被演算⼦に対して実⾏⼿続きを⽣成したのとそっ
くりです。シミュレーション時には、被演算⼦の⼿続きを呼び、演算をシミュ
レートする Scheme⼿続きをその結果の値に対して適⽤します。シミュレーシ
ョン⼿続きは、演算名でマシンの演算テーブルを検索することによって得られ
ます。
(define (lookup-prim symbol operations)

(let ((val (assoc symbol operations)))
(if val

(cadr val)
(error "Unknown operation: ASSEMBLE"

symbol))))

練習問題 5.9: マシンの演算を上のように扱うと、定数とレジスタ
の中⾝以外に、ラベルに対しても演算が可能になる。式処理⼿続
きを修正し、演算はレジスタと定数だけにしか使えないという条
件をチェックするようにせよ。
練習問題 5.10: レジスタマシンの命令に新しい構⽂を設計し、そ
の新しい構⽂を使えるようにシミュレータを修正せよ。この節の
構⽂⼿続き以外のシミュレータの部分を変更せずに新しい構⽂を
実装できるだろうか。
練習問題 5.11: 5.1.4節で saveと restoreを導⼊したときには、最
後に保存したレジスタでないレジスタを復元しようとしたときに
何が起こるかを規定していなかった。例えば、次のような命令列
の場合である。
(save y) (save x) (restore y)

restoreの意味として妥当なものはいくつか考えられる。
a. (restore y)は、どのレジスタから来た値か関係なしに、ス
タックに最後に保存された値を yに⼊れる。現在のシミュレ
ータのふるまいはこのようになっている。このふるまいを利

⽤して、5.1.4節のフィボナッチマシン (図 5.12)から命令を
ひとつ削減する⽅法を⽰せ。

b. (restore y)は、スタックに最後に保存された値が yを保存
したものである場合に限り、それを yに⼊れる。それ以外の
場合にはエラーを発する。シミュレータを修正し、ふるまい
をこのように変更せよ。スタックに値のほかにレジスタ名も
保存するように saveを変更する必要がある。

c. (restore y)は、最後に yから保存した値を yに⼊れる。y
の後にほかのレジスタの値が保存され、それらが取り出され
ていない場合にも、この挙動は変わらない。シミュレータを
修正し、ふるまいをこのように変更せよ。それぞれのレジス
タに対して、別々のスタックを関連づける必要があるだろう。
また、initialize-stack演算ではすべてのレジスタスタッ
クを初期化しなければならない。

練習問題 5.12: シミュレータは、あるコントローラを持つマシン
を実装するために必要となるデータパスを決めるときの補助とし
て使うことができる。アセンブラを拡張し、以下の情報をマシン
モデルに格納するようにせよ。

• 全命令のリスト。重複を削除し、命令の型 (assign, goto等)
によってソートしたもの。

• エントリポイントを⼊れるために使われるレジスタの (重複
なし)リスト (これらは goto命令によって参照されるレジス
タである)。

• saveまたは restoreされるレジスタの (重複なし)リスト。
• それぞれのレジスタに対する代⼊元の (重複なし)リスト (例
えば、図 5.11の階乗マシンでは、レジスタ val の代⼊元は
(const 1)と ((op *) (reg n) (reg val))である)。

マシンのメッセージパッシングインターフェイスを拡張し、この
新しい情報にアクセスできるようにせよ。この解析器をテストす
るため、図 5.12のフィボナッチマシンを定義し、構築されたリス
トを確認せよ。
練習問題 5.13: シミュレータを修正し、make-machineの引数とし
てレジスタリストを要求するのではなく、コントローラ命令列を使

ってマシンの持つレジスタを決定するようにせよ。make-machine
の中でレジスタをあらかじめ割り当てておく代わりに、命令アセ
ンブルの間に初めて現れるたびにレジスタを割り当てるようにし
てよい。

5.2.4 マシンパフォーマンスの監視
シミュレーションは、計画中のマシン設計の正しさを検証するのに役に⽴つだ
けでなく、マシンのパフォーマンスを測るためにも有⽤です。例えば、計算中
にスタック演算が使われた回数を測る “メーター”をシミュレータに組み込む
といったことができます。これを実現するためには、スタックをシミュレート
したものを修正し、スタックにレジスタが保存された回数とスタックが到達し
た最⼤の深さを管理するようにします。また、以下に⽰すように、スタックの
インターフェイスに統計情報を表⽰させるメッセージを追加します。さらに、
基本マシンモデルにスタック統計情報を⽰す演算を追加します。このために、
make-new-machineの the-opsを以下のように初期化します。
(list (list 'initialize-stack

(lambda () (stack 'initialize)))
(list 'print-stack-statistics

(lambda () (stack 'print-statistics))))

make-stackの新しいバージョンは以下のようになります。
(define (make-stack)

(let ((s '())
(number-pushes 0)
(max-depth 0)
(current-depth 0))

(define (push x)
(set! s (cons x s))
(set! number-pushes (+ 1 number-pushes))
(set! current-depth (+ 1 current-depth))
(set! max-depth (max current-depth max-depth)))

(define (pop)
(if (null? s)

(error "Empty stack: POP")
(let ((top (car s)))

(set! s (cdr s))
(set! current-depth (- current-depth 1))
top)))

(define (initialize)
(set! s '())
(set! number-pushes 0)
(set! max-depth 0)
(set! current-depth 0)
'done)

(define (print-statistics)
(newline)
(display (list 'total-pushes '= number-pushes

'maximum-depth '= max-depth)))
(define (dispatch message)

(cond ((eq? message 'push) push)
((eq? message 'pop) (pop))
((eq? message 'initialize) (initialize))
((eq? message 'print-statistics)
(print-statistics))

(else (error "Unknown request: STACK" message))))
dispatch))

練習問題 5.15から練習問題 5.19では、レジスタマシンシミュレータに追加でき
るこのほかの便利な監視・デバッグ機能について⽰します。

練習問題 5.14: 図 5.11に⽰した階乗マシンを使って、いくつかの
⼩さな nの値について n!の計算に必要となる pushの回数とスタ
ックの最⼤の深さを計測せよ。そのデータから、任意の n > 1に
ついて、n!を計算するのに使われる push演算の総数とスタック
の最⼤の深さを表す nの⽅程式を決定せよ。いずれも nの線形関
数であり、そのため⼆つの定数によって決定される。統計を表⽰
するためには、スタックを初期化する命令と統計を表⽰する命令
によって階乗マシンを拡張する必要がある。また、(図 5.4で gcd
マシンに対して⾏ったのと同じように)nの値を繰り返し読み込み、
階乗を計算し、結果を表⽰するようにマシンを修正してもいいか
もしれない。こうすると、get-register-contents,set-register-
contents!, startを何度も起動しないですむようになる。

練習問題 5.15: レジスタマシンシミュレータに命令数カウンタ
(instruction counting)を追加せよ。つまり、マシンモデルに実⾏
した命令数を管理させることになる。マシンモデルのインターフ
ェイスを拡張し、命令カウンタの値を表⽰してからカウンタをゼ
ロにリセットする新しいメッセージを受け⼊れるようにせよ。
練習問題 5.16: シミュレータを拡張し、命令トレース (instruction
tracing)を追加せよ。つまり、シミュレータが各命令を実⾏する前
に、その命令のテキストを表⽰するようにするということである。
マシンモデルが trace-on, trace-offメッセージを受け⼊れ、そ
れによってトレース機能をオン・オフするようにせよ。
練習問題 5.17: 練習問題 5.16の命令トレース機能を拡張し、シミ
ュレータが命令を表⽰する前に、コントローラ命令列中でその命
令の直前にあるラベルを表⽰するにせよ。これを⾏う際に、命令
数カウンタ (練習問題 5.15)に⼲渉しないように注意すること。必
要なラベル情報をシミュレータに持たせておく必要があるだろう。
練習問題 5.18: 5.2.1節の make-register⼿続きを修正し、レジス
タをトレースできるようにせよ。レジスタはトレースをオン・オ
フするメッセージを受け⼊れる。レジスタがトレースされている
ときには、そのレジスタへの代⼊が⾏われると、レジスタの名前
・レジスタの古い値・代⼊する新しい値が表⽰される。マシンモ
デルのインターフェイスを拡張し、指定されたマシンレジスタの
トレース機能をオン・オフできるようにせよ。
練習問題 5.19: Alyssa P. Hackerは、マシン設計のデバッグに使う
ために、シミュレータにブレークポイント (breakpoint)機能がほ
しくなった。そこで、この機能を組み込むために、彼⼥はあなたを
雇うことにした。彼⼥は、シミュレータが⽌まる場所を指定できる
ようにして、またマシンの状態を調べられるようにしてほしいと
思っている。あなたは以下の⼿続きを実装することになっている。
(set-breakpoint ⟨machine⟩ ⟨label⟩ ⟨n⟩)

この⼿続きは、与えられたラベルから n番⽬の命令の直前にブレ
ークポイントを設定するというものだ。例えば、
(set-breakpoint gcd-machine 'test-b 4)

のようにすると gcd-machineでレジスタ aへの代⼊の直前にブレ
ークポイントが設定される。シミュレータがブレークポイントに
到達すると、ラベルとブレークポイントのオフセットを表⽰し、命
令の実⾏を停⽌する。すると、Alyssaは get-register-contents
と set-register-contents!を使ってシミュレートされているマ
シンの状態を操作できるようになる。また、Alyssaがその後以下
のように⼊⼒することによって実⾏を継続できるようにしなけれ
ばならない。
(proceed-machine ⟨machine⟩)

さらに、以下のようにして特定のブレークポイントを削除したり、
(cancel-breakpoint ⟨machine⟩ ⟨label⟩ ⟨n⟩)

以下のようにしてすべてのブレークポイントを削除したりできる
ようにしなければならない。
(cancel-all-breakpoints ⟨machine⟩)

5.3 記憶領域の割り当てとガベージコレクショ
ン

5.4節では、Scheme評価器をレジスタマシンとして実装する⽅法を⽰します。考
察を簡単にするために、このレジスタマシンはリスト構造メモリ (list-structured
memory)を備えていて、リスト構造のデータを操作する演算は基本演算だとし
ます。そのようなメモリの存在を仮定するというのは、Schemeインタプリタ
の制御メカニズムに焦点を当てている場合には役に⽴つ抽象化ですが、これは
現代のコンピュータにおける実際の基本データ演算の現実的な⾒⽅を反映して
いません。Lispシステムがどのように動いているかということについてより全
⾯的な理解を得るためには、従来のコンピュータメモリで実現できる形でリス
ト構造を表現する⽅法を調査する必要があります。
リスト構造の実装にあたっては、考えるべきことが⼆つあります。⼀つ⽬

は、純粋に表現の問題です。Lispペアの “箱とポインタ”という構造を、典型
的なコンピュータメモリの記憶領域とアドレス機能だけを使って表現するには
どのようにすればよいかということです。⼆つ⽬の問題は、計算の進⾏に伴う

メモリ管理に関するものです。Lispシステムの演算は、次々と新しいデータオ
ブジェクトを作る能⼒に強く依存しています。こうして作られるデータオブジ
ェクトには、Lispの⼿続きが解釈される間に明⽰的に作られるもののほかに、
環境や引数リストといったインタプリタ⾃⾝によって作られる構造も含まれま
す。絶えず新しいデータオブジェクトを作っていくということは、⾼速にアク
セスできる無限のメモリを持つコンピュータであれば何の問題にもならないと
ころですが、実際にはコンピュータのメモリは有限です (残念ながら)。そのた
め、Lispシステムは⾃動記憶領域割り当て (automatic storage allocation) とい
う機能を⽤意して、メモリが無限であるかのような幻想を作り出しています。
データオブジェクトが必要でなくなると、それに割り当てられていたメモリは
⾃動的にリサイクルされ、新しいデータオブジェクトを構築するのに使われま
す。そのような⾃動記憶領域割り当てを提供する技術はいくつもあります。こ
の節で検討する⼿法は ガベージコレクション (garbage collection)と呼ばれる
ものです。

5.3.1 ベクタとしてのメモリ
従来型コンピュータのメモリは、情報をしまっておける⼀列の枠を持つ、棚
のように考えることができます。それぞれの枠は、アドレス (address)や 位置
(location)と呼ばれる⼀意の名前を持っています。普通のメモリシステムは⼆
つの基本演算を提供します。ひとつは指定された位置に格納されたデータを取
ってくる演算で、もうひとつは指定された位置に新しいデータをしまう演算で
す。枠の集合に順にアクセスできるように、メモリアドレスはインクリメント
できるようになっています。より⼀般的に、多くの重要なデータ演算では、メ
モリアドレスをデータとして、メモリ位置に格納したり、マシンのレジスタで
操作したりできる必要があります。リスト構造の表現は、そのようなアドレス
演算 (address arithmetic)の応⽤のひとつです。
コンピュータメモリをモデル化するために、ここではベクタ (vector)と呼

ばれる新しい種類のデータ構造を使います。抽象的には、ベクタは整数のイン
デックスによって個々の要素にアクセスでき、またそのアクセス時間がインデ
ックスとは独⽴であるような複合データ構造です。5 メモリ演算を記述するに
は、ベクタを操作する Schemeの基本⼿続き⼆つを使います。

• (vector-ref <vector> <n>)は、ベクタの n番⽬の要素を返します。
5メモリを項⽬のリストとして表現することも可能です。しかし、その場合アクセス時

間はインデックスとは独⽴でなくなってしまいます。リストの n番⽬の要素にアクセス
するには、n− 1回の cdr演算が必要だからです。

• (vector-set! <vector> <n> <value>)はベクタの n番⽬の要素に指定
された値を設定します。

例えば、vがベクタであれば、(vector-ref v 5)はベクタ vの 5番⽬の項⽬を
取得し、(vector-set! v 5 7)はベクタ vの 5番⽬の項⽬の値を 7に変更しま
す。6 コンピュータメモリでは、メモリ内でのベクタの開始位置を指定するベ
ースアドレス (base address)とベクタの特定の項⽬のオフセットを指定するイ
ンデックス (index)をアドレス演算を使って組み合わせることによってこのア
クセスを実装できます。

Lispデータの表現
リスト構造メモリのために必要となる基本的なペア構造は、ベクタを使って実
装できます。コンピュータメモリが the-carsと the-cdrsという⼆つのベク
タに分かれていると想像してみましょう。リスト構造は、以下のように表現す
ることにします。まず、ペアへのポインタはこの⼆つのベクタのインデックス
とします。ペアの carは the-carsの指定されたインデックスの項⽬で、ペア
の cdrは the-cdrsの指定されたインデックスの項⽬です。また、ペア以外の
オブジェクト (数値や記号など)の表現と、それぞれのデータの種類を判別する
⽅法も必要になります。これを達成する⼿法はいろいろありますが、いずれも
結局は 型つきポインタ (typed pointer)を使うというところに帰着します。型
つきポインタというのは、“ポインタ”の概念を拡張し、データ型の情報を持た
せるようにするということです。7 データ型があると、ペア (“ペア”というデ
ータ型と、メモリベクタのインデックスからなる) のポインタと、ほかの種類
のデータ (ほかの何らかのデータ型と、その型を表現するのに使われる何らか
のもの)のポインタをシステムが判別できるようになります。⼆つのデータオ
ブジェクトは、それらのポインタが同⼀である場合に同じ (eq?)であると⾒な
されます。8 図 5.14は、この⼿法によってリスト ((1 2) 3 4)を表現した図で、
箱とポインタの図も同時に⽰しています。ここでは、前につけた⽂字によって

6ベクタを完全なものにするためには、ベクタを構築する make-vector演算を規定し
なければならないところです。しかし、ここでのアプリケーションでは、コンピュータ
メモリの決まった部分をモデル化するためだけにしか使いません。

7これは、第 2章でジェネリック演算を扱うために導⼊した “タグつきデータ”とまっ
たく同じ考え⽅です。しかし、ここではデータ型をリストを使って構築するのではなく、
マシンの基本レベルで含めるようにしています。

8型情報は Lispシステムを実装するマシンがどのようなものであるかによっていろい
ろな符号化⽅法が考えられます。Lispプログラムの実⾏効率は、この選択をどれだけう
まくやるかに強く依存します。しかし、どのような選択をするのがよいかという⼀般的

データ型情報を⽰しています。つまり、インデックス 5を持つペアへのポイン
タは p5、空リストはポインタ e0、数値 4へのポインタは n4によって⽰されま
す。箱とポインタの図では、それぞれのペアの左下に、そのペアの carと cdr
が格納される位置を⽰すベクタのインデックスを記しています。the-carsと
the-cdrsの空⽩の場所には、(ここで扱っていない)ほかのリスト構造の⼀部
などが⼊っている可能性もあります。
数値へのポインタ、例えば n4は、数値データを⽰す型フィールドと、数値

4を実際に表現したデータからなることもあります。9 ひとつのポインタに割
り当てられる固定⻑のメモリでは表現できないほど⼤きい数値を扱うには、別
に多倍⻑整数 (bignum)データ型を使うことが考えられます。この場合、ポイ
ンタは数値の各部分を格納したリストを指すことになります。10

記号は、記号の⽂字表現を構成する⽂字の列を指す型つきポインタとして
表現することが考えられます。この列は、⼊⼒の中で⽂字列が最初に出てきた
ときに Lispの読み込み器によって構築されます。ここで、ひとつの記号が⼆箇
所に出てきたとき、それらは eq?によって “同じ”記号として認識されるよう
にしたいところです。それと同時に、eq?はポインタの等価性の簡単なテスト
としておきたいところです。そのため、読み込み器が同じ⽂字列を⼆回⾒た場
合には、その⼆回の出現両⽅に対して同じポインタ (同じ⽂字列を指すポイン
タ)を使うということを保証する必要があります。これを達成するために、読
み込み器は出てきた記号すべてのテーブルを管理します。このテーブルは伝統
的にオブアレイ (obarray)と呼ばれます。読み込み器が⽂字列にぶつかって記
号を構築しようとするとき、同じ⽂字列がそれまでに出てきていないかオブア
な設計規則を定式化することは簡単ではありません。型つきポインタを実装する最も素
直なやり⽅は、ポインタ中の何ビットかを、データ型を表す型フィールド (type field)と
して使うというものです。このような表現を設計するにあたっては、考慮すべき重要な
問題がいくつかあります。型ビットは何ビット必要か、ベクタのインデックスはどれだ
け必要か、ポインタの型フィールドを操作するのにマシンの基本命令がどれだけ効率的
に使えるか、といったものです。型フィールドを効率よく扱うための特別なハードウェ
アを持つマシンは タグアーキテクチャ (tagged architecture)を持つというように⾔われ
ます。

9ここで数値をどう表現することにするかによって、ポインタの等価性をテストする
eq?を数値の等値性のテストに使えるかどうかが決まります。ポインタの値が数値デー
タそのものを含んでいれば、等しい数値はポインタとしても同じ値を持つことになりま
す。しかし、ポインタが数値を格納している位置へのインデックスを持っているのであ
れば、同じ数値が複数の位置に格納されることが決してないようにしない限り、等しい
数値が同じポインタ値になることは保証されません。

10これは数値を数字の列として書くのと似ていますが、この場合、各 “桁”は 0からひ
とつのポインタに格納できる最⼤値の間の数値となります。

図 5.14: リスト ((1 2) 3 4)の箱とポインタ表現とメモリ
ベクタ表現

レイをチェックします。出てきていなければ、その⽂字列を使って新しい記号
(新しい⽂字列を指す型つきポインタ)を構築し、そのポインタをオブアレイに
⼊れます。読み込み器がその⽂字列をすでに⾒たことがあれば、読み込み器は
オブアレイに格納されている記号ポインタを返します。⽂字列を⼀意なポイン
タで置き換えるこの処理は、記号の閉じ込め (interning)と呼ばれます。

基本リスト演算の実装
ここまでに述べたような表現法を使うと、レジスタマシンのそれぞれの “基
本” リスト命令をいくつかの基本ベクタ演算で置き換えられるようになりま
す。ここではメモリベクタを指す⼆つのレジスタ the-carsと the-cdrsを使
い、vector-refと vector-set!は基本演算として使えると仮定します。また、
ポインタ演算 (ポインタのインクリメント、ペアポインタによるベクタの位置
指定、⼆つの数値の加算など)は、型つきポインタのインデックス部だけを使
うものと仮定します。
例えば、以下の命令をサポートするレジスタマシンは、

(assign ⟨reg1⟩ (op car) (reg ⟨reg2⟩))
(assign ⟨reg1⟩ (op cdr) (reg ⟨reg2⟩))

それぞれ次のように実装することによって作ることができます。
(assign ⟨reg1⟩ (op vector-ref) (reg the-cars) (reg ⟨reg2⟩))
(assign ⟨reg1⟩ (op vector-ref) (reg the-cdrs) (reg ⟨reg2⟩))

以下の命令は、
(perform (op set-car!) (reg ⟨reg1⟩) (reg ⟨reg2⟩))
(perform (op set-cdr!) (reg ⟨reg1⟩) (reg ⟨reg2⟩))

次のように実装できます。
(perform
(op vector-set!) (reg the-cars) (reg ⟨reg1⟩) (reg ⟨reg2⟩))

(perform
(op vector-set!) (reg the-cdrs) (reg ⟨reg1⟩) (reg ⟨reg2⟩))

consは、未使⽤のインデックスを割り当て、the-carsと the-cdrsの中のその
インデックスが指すベクタ位置に consの⼆つの引数を格納することによって
実⾏できます。ここでは、次に使⽤可能なインデックスの⼊ったペアポインタ
を持つ freeという特別なレジスタが存在し、そのポインタのインデックス部
をインクリメントすることで次の空き位置を⾒つけられると仮定します。11 例
えば、次の命令は、
(assign ⟨reg1⟩ (op cons) (reg ⟨reg2⟩) (reg ⟨reg3⟩))

以下のようなベクタ命令の列として実装できます。12

(perform
(op vector-set!) (reg the-cars) (reg free) (reg ⟨reg2⟩))

(perform
(op vector-set!) (reg the-cdrs) (reg free) (reg ⟨reg3⟩))

11空き領域を探す⽅法はほかにもあります。例えば、未使⽤のペアをすべて 空きリス
ト (free list)として連結するということも可能です。5.3.2節で⾒ていくように、ここで
は圧縮 GCを使うため、空き位置は連続的 (そのため、ポインタのインクリメントによ
ってアクセス可能)です。

12これは、本質的には3.3.1節で述べた set-car!と set-cdr!による consの実装と同
じです。そのときの実装で使った get-new-pair演算は、ここでは freeポインタによっ
て実現されています。

(assign ⟨reg1⟩ (reg free))
(assign free (op +) (reg free) (const 1))

以下の eq?演算は、
(op eq?) (reg ⟨reg1⟩) (reg ⟨reg2⟩)

単純にレジスタのすべてのフィールドの等価性をテストします。pair?, null?,
symbol?, number?のような述語は、型フィールドをチェックするだけで⼗分
です。

スタックの実装
私たちのレジスタマシンはスタックを使いますが、スタックはリストによって
モデル化できるため、ここでは特別なことをする必要はありません。スタック
は保存した値のリストとして実装できます。the-stackという特別なレジスタ
がスタックを指すものとします。そのため、(save <reg>)は次のように実装
できます。
(assign the-stack (op cons) (reg ⟨reg⟩) (reg the-stack))

同じように、(restore <reg>)は次のように実装できます。
(assign ⟨reg⟩ (op car) (reg the-stack))
(assign the-stack (op cdr) (reg the-stack))

また、(perform (op initialize-stack))は次のように実装できます。
(assign the-stack (const ()))

これらの演算は、上で⽰したベクタ演算によってさらに展開できます。なお、
従来型コンピュータアーキテクチャでは、スタックを別のベクタとして割り当
てるほうが通常は有利です。その場合、プッシュとポップはそのベクタのイン
デックスのインクリメントとデクリメントとして実現できます。

練習問題 5.20: freeポインタの初期値を p1として、次の式によっ
て⽣成されるリスト構造について、(図 5.14のように)箱とポイン
タ表現とメモリベクタ表現を描け。
(define x (cons 1 2))
(define y (list x x))

freeの最終的な値は何になるか。どのポインタが xと yの値を表
現しているだろうか。
練習問題 5.21: 以下の⼿続きに対応するレジスタマシンを実装せ
よ。リスト構造メモリ演算はマシンの基本演算として使えると仮
定する。

a. 再帰的 count-leaves:

(define (count-leaves tree)
(cond ((null? tree) 0)

((not (pair? tree)) 1)
(else (+ (count-leaves (car tree))

(count-leaves (cdr tree))))))

b. 明⽰的にカウンタを持つ再帰的 count-leaves

(define (count-leaves tree)
(define (count-iter tree n)

(cond ((null? tree) n)
((not (pair? tree)) (+ n 1))
(else
(count-iter (cdr tree)

(count-iter (car tree)
n)))))

(count-iter tree 0))

練習問題 5.22: 3.3.1節の練習問題 3.12では、⼆つのリストを⾜し
てひとつの新しいリストにする append⼿続きと、⼆つのリストを
継ぎ合わせる append!⼿続きを紹介した。これらの⼿続きそれぞ
れを実装するレジスタマシンを設計せよ。リスト構造メモリ演算
は基本演算として使えると仮定する。

5.3.2 無限メモリ幻想の維持
5.3.1節で概観した表現⼿法は、無限の容量のメモリさえあるなら、リスト構
造の実装という問題の解決になっています。しかし、実際のコンピュータでは、

新しいペアを構築するための空き領域をそのうち使い果たしてしまいます。13

しかし、典型的な計算で⽣成されるペアのほとんどは中間結果を保存するため
だけに使われるものです。結果にアクセスされた後は、それらのペアはもう必
要なくなります—つまり、ゴミ (garbage)になるということです。例えば、次
の計算は、
(accumulate + 0 (filter odd? (enumerate-interval 0 n)))

列挙リストと、列挙リストをフィルタリングした結果のリストという⼆つを構
築します。集積が完了すると、これらのリストはもう必要なく、割り当てたメ
モリは再利⽤できます。すべてのゴミを定期的に回収する仕組みを作り、それ
によってメモリを再利⽤する速度が新しいペアを構築するのとだいたい同じ
速度であれば、無限の容量のメモリがあるという幻想を維持できたことになり
ます。
ペアを再利⽤するには、どの割り当て済みペアが (ペアの中⾝が将来の計

算に影響することがないという意味で)不要かを決定する⽅法が必要です。こ
れを達成するためにここで検討する⼿法は、ガベージコレクション (garbage
collection)として知られるものです。ガベージコレクションの基本となる事実
は、Lispの逐次実⾏のどの時点でも、将来の計算に影響を与えうるオブジェク
トは、現在マシンのレジスタ内にあるポインタから carと cdrを繰り返して到
達できるものだけだということです。14 そのようなアクセスが可能でないメモ
リセルは再利⽤できます。
ガベージコレクションを実⾏する⽅法はいろいろあります。ここで検討す

る⼿法はストップ&コピー (stop-and-copy)と呼ばれるものです。基本的な考
え⽅は、メモリを “ワーキングメモリ”と “フリーメモリ”の⼆つに分割すると
いうものです。consがペアを構築するときは、ペアをワーキングメモリに割り

13これが当てはまらなくなるときが来る可能性はあります。コンピュータの耐⽤期間
中に空き領域を使い切れないほどメモリが⼤きくなるということも考えられるからです。
例えば、⼀年は 3 · 1013マイクロ秒なので、1マイクロ秒に 1回 consを⾏うとすると、メ
モリを使い切ることなしに 30年間運⽤可能なコンピュータを作るのには約 1015 のメモ
リセルが必要ということになります。それだけの容量のメモリというのは、今の尺度で
考えると⾺⿅⾺⿅しいほど巨⼤なものに思えますが、物理的に不可能というわけではあ
りません。⼀⽅、プロセッサは速くなってきていますし、未来のコンピュータはひとつの
メモリに対して並列に演算を⾏う多数のプロセッサを持つようになるかもしれないので、
ここでの想定よりもずっと速くメモリを使い果たすようになることも考えられます。

14ここでは、スタックは5.3.1節で書いたようにリストとして表現されていると仮定し
ます。そのため、スタック上の項⽬はスタックレジスタのポインタを経由してアクセス
可能ということになります。

当てます。ワーキングメモリがいっぱいになるとガベージコレクションを⾏い
ます。ガベージコレクションは、ワーキングメモリ内の有⽤なペアをすべて探
し、それらをフリーメモリの連続した位置にコピーすることによって⾏います。
(有⽤なペアは、マシンレジスタからすべての carと cdrのポインタをたどっ
ていくことによって探します)。ゴミはコピーしないので、新しいペアを割り当
てるのに使える余分のフリーメモリがおそらくできているはずです。それに加
えて、ワーキングメモリの中⾝はすべて必要ありません。有⽤なペアはすべて
コピー済みだからです。つまり、ワーキングメモリとフリーメモリの役割を交
換すると、処理を続けられることになります。新しいペアは新しいワーキング
メモリ (元フリーメモリ)に割り当てられ、それがいっぱいになったなら、有⽤
なペアを新しいフリーメモリ (元ワーキングメモリ)にコピーできます。15

ストップ &コピーガベージコレクタの実装
ここからは、レジスタマシン⾔語を使ってストップ &コピーアルゴリズムを
さらに詳しく⾒ていきます。ここでは、rootというレジスタがあり、そのレジ
スタはある構造へのポインタを含んでいて、アクセス可能なデータはそのポイ

15この考え⽅は Minskyによって発明され、実装されたものです。Minskyの実装は、
mit研究所の電⼦⼯学研究所にあった pdp-1 で動かす Lisp 実装の⼀部でした。これは、
Multics時分割システムの Lisp実装で使うために、Fenichel and Yochelson (1969)によ
ってさらに開発が進められました。のちにBaker (1978)は、ガベージコレクションの間
に計算を⽌める必要がない、この⼿法の “リアルタイム”版を開発しました。Bakerの
考え⽅は Hewitt, Lieberman, Moonによって拡張され (Lieberman and Hewitt 1983参
照)、データ構造には揮発性のものもあればより永続的なものもあるということを利⽤す
るようになりました。
このほかによく使われるガベージコレクション技術としては、マーク & スイープ

(mark-sweep)⼿法があります。これは、マシンレジスタからアクセス可能な構造をすべ
てたどり、到達できたペアそれぞれに印をつける (マークする)というものです。それか
らメモリ全部を⾛査し、印のついていないメモリをすべてゴミとして “掃き出し (スイー
プし)”、再使⽤可能にします。マーク &スイープについては、Allen 1978で詳細に検討
されています。

Minsky-Fenichel-Yochelsonアルゴリズムはメモリの有⽤な部分のみを調査するため、
⼤きなメモリを持つシステムでは主流となっています。これは、スイープの段階でメモリ
すべてをチェックする必要があるマーク&スイープとは対照的です。ストップ&コピー
のもうひとつの利点は圧縮 (compacting)ガベージコレクタであるというところです。こ
れは、ガベージコレクション段階が終わったときには有⽤なデータは連続したメモリ位
置に移動させられていて、ゴミとなるペアは排除されているということです。このこと
は、広く散らばったメモリアドレスにアクセスすると余計なページング処理が必要とな
る仮想メモリを持つ機械では、⾮常に重要なパフォーマンス上の考慮点になりえます。

ンタからすべてたどれると仮定します。この条件を整えるには、ガベージコレ
クションを⾏う直前にすべてのマシンレジスタの中⾝を事前に割り当てたリス
トに格納し、rootがそれを指すようにします。16 さらに、現在のワーキング
メモリのほかに、有⽤なデータをコピーできるフリーメモリがあると仮定しま
す。現在のワーキングメモリは the-carsと the-cdrsというレジスタにベース
アドレスが⼊っているベクタからなり、フリーメモリは new-carsと new-cdrs
というレジスタにベースアドレスが⼊っているベクタからなります。
ガベージコレクションは、現在のワーキングメモリ内の空きセルがなくな

ったときに起動されます。つまり、cons演算がメモリベクタの終端を超えて
freeポインタをインクリメントしようとしたときということです。ガベージ
コレクション処理が終わったときには、rootポインタは新しいメモリを指し、
rootからアクセス可能なオブジェクトはすべて新しいメモリに移動されてい
ます。また、freeポインタは新しいメモリの中で新しいペアを割り当てること
ができる次の位置を⽰しています。それに加えて、ワーキングメモリと新しい
メモリの役割が交換されています。新しいペアは freeの指す位置から始まる
新しいメモリの中に構築され、(以前の)ワーキングメモリは次のガベージコレ
クションのための新しいメモリとして利⽤可能になっています。図 5.15はガベ
ージコレクション直前と直後のメモリの配置を⽰したものです。
ガベージコレクション処理の状態は、freeと scanという⼆つのポインタ

を管理することによってコントロールされています。これらのレジスタは、新
しいメモリの先頭を指すように初期化されます。まず、このアルゴリズムは
rootの指すペアを新しいメモリの先頭に再配置します。そのペアはコピーさ
れ、rootポインタはインクリメントされます。それに加えて、ペアの古い位置
にはその中⾝が移動されたことを⽰す印をつけます。これは次のように⾏いま
す。carの位置には、これが移動済みオブジェクトであることを⽰す特別なタ
グを置きます (そのようなオブジェクトは伝統的に 失恋ペア (broken heart)と
呼ばれます)。17 cdrの位置には、オブジェクトの移動先の位置を指す移転先ア
ドレス (forwarding address)を置きます。

rootを再配置した後は、ガベージコレクタは基本サイクルに⼊ります。ア
ルゴリズムの各ステップで、scan ポインタ (最初は再配置後の root を指す)
は、新しいメモリに移動されたものの carポインタと cdrのポインタが依然と
して古いメモリ内のオブジェクトを参照しているペアを指しています。それら

16このレジスタリストは、記憶領域割り当てシステムが使うレジスタ—root, the-
cars, the-cdrsや、この節で紹介するほかのレジスタは含みません。

17失恋ペアという⽤語はDavid Cresseyによる造語です。彼は 1970年代初期に mitで
開発された Lisp⽅⾔であるMDLのガベージコレクタを書きました。

図 5.15: ガベージコレクション処理によるメモリの再構成

のオブジェクトはそれぞれ再配置され、scanポインタはインクリメントされ
ます。オブジェクト (例えば、⾛査しているペアの carポインタの指すオブジ
ェクト)を再配置するときは、そのオブジェクトが移動済みかどうか (そのオブ
ジェクトの carの位置に失恋タグがあるかどうか)を確認します。オブジェク
トが移動済みでなければ、freeによって⽰される位置にそのオブジェクトを
コピーして、freeを更新し、オブジェクトの古い位置に失恋ペアを設定し、そ
のオブジェクトへのポインタ (この例では、⾛査中のペアの carポインタ)がオ
ブジェクトの新しい位置を指すように更新します。オブジェクトが移動済みで
ある場合は、⾛査中のペアのポインタを、(失恋ペアの cdrの位置にある)移転
先に置き換えます。最終的に、アクセス可能なオブジェクトすべての移動と⾛
査が終わります。そのとき scanポインタは freeポインタに追いつき、処理は
停⽌します。
ストップ &コピーアルゴリズムはレジスタマシンの命令列として規定でき

ます。オブジェクト再配置の基本ステップは relocate-old-result-in-newと
呼ばれるサブルーチンによって実⾏されます。このサブルーチンは、再配置対
象オブジェクトのポインタという引数を oldという名前のレジスタから取得し
ます。サブルーチンは指定されたオブジェクトを再配置し (その過程で freeを
インクリメントし)、再配置済みオブジェクトへのポインタを newと呼ばれる
レジスタに⼊れてリターンします。リターンは、relocate-continueレジスタ
に格納されたエントリポイントに分岐することによって⾏います。ガベージコ
レクションを始めるためには、freeと scanを初期化してからこのサブルーチ
ンを起動して rootポインタを再配置します。rootの再配置が終わったら、new
ポインタを新しい rootとして設定し、ガベージコレクタのメインループに⼊
ります。
begin-garbage-collection

(assign free (const 0))
(assign scan (const 0))
(assign old (reg root))
(assign relocate-continue (label reassign-root))
(goto (label relocate-old-result-in-new))

reassign-root
(assign root (reg new))
(goto (label gc-loop))

ガベージコレクタのメインループでは、⾛査するべきオブジェクトがまだある
かどうか決定する必要があります。そのためには、scanポインタと freeポイ

594



ンタが同じ位置にあるかどうかをテストします。ポインタが等しければ、アク
セス可能なオブジェクトはすべて再配置が終わっていることになるので、gc-
flipに分岐します。gc-flipでは、中断された計算を継続できるように後⽚付
けを⾏います。まだ⾛査するべきペアが残っていれば、(carポインタを oldに
⼊れることによって) 再配置サブルーチンを呼び出して次のペアの carを再配
置します。サブルーチンからリターンした後に carポインタを更新するように
relocate-continueレジスタを設定しておきます。
gc-loop

(test (op =) (reg scan) (reg free))
(branch (label gc-flip))
(assign old (op vector-ref) (reg new-cars) (reg scan))
(assign relocate-continue (label update-car))
(goto (label relocate-old-result-in-new))

update-carでは、⾛査中のペアの carポインタを修正し、次にペアの cdrを
再配置するために向かいます。再配置が完了すると update-cdrにリターンし
ます。再配置と cdrの更新が終わったらそのペアの⾛査は終わりなので、メイ
ンループを続けます。
update-car

(perform (op vector-set!)
(reg new-cars)
(reg scan)
(reg new))

(assign old (op vector-ref) (reg new-cdrs) (reg scan))
(assign relocate-continue (label update-cdr))
(goto (label relocate-old-result-in-new))

update-cdr
(perform (op vector-set!)

(reg new-cdrs)
(reg scan)
(reg new))

(assign scan (op +) (reg scan) (const 1))
(goto (label gc-loop))

サブルーチン relocate-old-result-in-newは、次のようにしてオブジェクト
を再配置します。まず、(oldの指す)再配置対象のオブジェクトがペアでない

なら、そのオブジェクトへの同じポインタをそのまま (newの中で)返します
(例えば、carが数値の 4であるペアを⾛査しているとします。ここで、5.3.1
節で記述したようにその carを n4として表現しているとすると、“再配置済
み”の carのポインタも、そのまま n4であってほしいところです)。ペアであれ
ば、再配置を⾏う必要があります。もし再配置対象のペアの carの位置に失恋
ペアのタグがあるなら、そのペアは実際にはもう移動済みなので、(失恋ペアの
cdrの位置から)移転先を取得し、それを newに⼊れて返します。old内のポイ
ンタが移動済みでないペアを指している場合は、そのペアを (freeの指す)新
しいメモリの最初の空きセルに移動させ、元の位置には失恋タグと移動先を格
納して失恋ペアを設定します。relocate-old-result-in-newは、oldの指す
オブジェクトの carまたは cdrを⼊れるためにレジスタ oldcrを使います。18

relocate-old-result-in-new
(test (op pointer-to-pair?) (reg old))
(branch (label pair))
(assign new (reg old))
(goto (reg relocate-continue))

pair
(assign oldcr (op vector-ref) (reg the-cars) (reg old))
(test (op broken-heart?) (reg oldcr))
(branch (label already-moved))
(assign new (reg free)) ; ペアの新しい位置
;; freeポインタを更新する
(assign free (op +) (reg free) (const 1))
;; carと cdrを新しいメモリにコピー
(perform (op vector-set!)

(reg new-cars) (reg new) (reg oldcr))
(assign oldcr (op vector-ref) (reg the-cdrs) (reg old))
(perform (op vector-set!)

(reg new-cdrs) (reg new) (reg oldcr))
;; 失恋ペアの構築

18ガベージコレクタでは、リスト構造の pair?演算ではなく、低レベル述語の pointer-
to-pair?を使います。これは、実際のシステムではさまざまなデータを、ガベージコ
レクションのときだけペアとして扱いたいことがあるからです。例えば、ieee標準に準
拠した Scheme システムでは、⼿続きオブジェクトは述語 pair? を満たさない特別な
種類の “ペア” として実装することが許されています。ここでのシミュレーションでは、
pointer-to-pair?は pair?として実装しても問題ありません。

(perform (op vector-set!)
(reg the-cars) (reg old) (const broken-heart))

(perform
(op vector-set!) (reg the-cdrs) (reg old) (reg new))

(goto (reg relocate-continue))
already-moved

(assign new (op vector-ref) (reg the-cdrs) (reg old))
(goto (reg relocate-continue))

ガベージコレクション処理を終える前に、ポインタを交換して古いメモリと新
しいメモリの役割を交換します。交換するのは the-cars と new-cars、the-
cdrsと new-cdrsです。これで、次にメモリを使い切ったときにまたガベージ
コレクションを実⾏するための準備ができたことになります。
gc-flip

(assign temp (reg the-cdrs))
(assign the-cdrs (reg new-cdrs))
(assign new-cdrs (reg temp))
(assign temp (reg the-cars))
(assign the-cars (reg new-cars))
(assign new-cars (reg temp))

5.4 明⽰制御評価器
5.1節では、簡単な Schemeプログラムをレジスタマシンの記述に変形する⽅
法を⾒てきました。ここでは、この変形をより複雑なプログラムである4.1.1
節–4.1.4節のメタ循環評価器について⾏います。メタ循環評価器は、Schemeイ
ンタプリタのふるまいを eval⼿続きと apply⼿続きによって記述する⽅法を
⽰しています。この節で開発する 明⽰制御評価器 (explicit-control evaluator)
は、評価処理の基層として使われている⼿続き呼び出しと引数渡しのメカニズ
ムをレジスタとスタックの演算によって記述する⽅法を⽰します。さらに、明⽰
制御評価器は従来型コンピュータの機械語によく似た⾔語で書かれた Scheme
インタプリタの実装でもあります。この評価器は5.2節のレジスタマシンシミ
ュレータで実⾏できます。Scheme評価器の機械語による実装を作るための最
初の⼀歩としてこの評価器を使うこともできますし、Schemeの式を評価する
特殊⽤途マシンを作るうえでの⾜がかりとすることさえ可能です。図 5.16にそ

図 5.16: Scheme評価器のシリコンチップ実装

のようなハードウェアによる実装を⽰します。これは、Schemeの評価器とし
て動作するシリコンチップです。チップの設計者は、この節で記述する評価器
に似たレジスタマシンのデータパスとコントローラの仕様からスタートし、集
積回路のレイアウトの構築には設計⾃動化プログラムを使っています。19

レジスタと演算
明⽰制御評価器を設計するにあたっては、レジスタマシンで使う演算を規定す
る必要があります。メタ循環評価器は quoted?や make-procedureのような⼿
続きを使って抽象構⽂によって記述しました。レジスタマシンを実装する際に
は、これらの⼿続きを基本的なリスト構造メモリ演算の列として展開し、これ
らの演算をレジスタマシン上で実装することも可能です。しかし、そうすると
評価器はとても⻑くなり、基本的な構造が実装の細部に紛れてわかりにくくな

19このチップとその設計⼿法についての詳細はBatali et al. 1982を参照。

ってしまいます。⾒た⽬をすっきりさせるために、4.1.2節で出てきた構⽂⼿続
きや、4.1.3節と4.1.4節で出てきた環境その他の実⾏時データを表現する⼿続
きはレジスタマシンの基本演算に含めることにします。低レベルの機械語によ
るプログラムやハードウェアでの実装が可能な評価器を完全に規定するために
は、5.3節に記述したリスト構造の実装を使って、より基本的な演算によって
これらの演算を置き換えることになります。
ここで作る Scheme評価器レジスタマシンは、ひとつのスタックと七個の

レジスタ、exp, env, val, continue, proc, argl, unevを持っています。expは
評価対象の式を保持し、envは評価を⾏う環境を保持します。評価終了時には、
指定された環境で式を評価することによって得られた結果が valに⼊ります。
continueレジスタは、5.1.4節で説明したように再帰の実装に使います (評価
器は⾃分⾃⾝を再帰的に呼び出す必要があります。式の評価の際には、その部
分式を評価することが必要になるからです)。レジスタ proc, argl, unevは組
み合わせの評価に使います。
ここでは、評価器のレジスタと演算の接続をデータパス図を描いて⽰すこ

とはしません。また、マシンの演算の完全なリストも提供しません。それらは
評価器のコントローラから間接的にわかるからです。コントローラについては
詳細に解説します。

5.4.1 明⽰制御評価器のコア
評価器の中⼼となる要素は、eval-dispatchから始まる命令列です。これは4.1.1
節で記述したメタ循環評価器の eval⼿続きに対応するものです。コントロー
ラが eval-dispatch からスタートすると、コントローラは expによって指定
される式を envによって指定される環境の中で評価します。評価が完了すると、
コントローラは continueに格納されたエントリポイントに⾏きます。そのと
き、valレジスタには式の値が⼊っています。メタ循環評価器の evalと同じよ
うに、eval-dispatchは、評価対象の式の構⽂型に対する場合分けという構造
になっています。20

eval-dispatch

20このコントローラでは、ディスパッチは testと branchの命令列として書かれてい
ます。別のやり⽅として、テストを連続して実⾏することを避けるため、また新しい式
の型の定義をやりやすくするために、ディスパッチをデータ主導スタイルで書くことも
考えられます (また、実際のシステムではおそらくそうなっています)。Lisp実⾏のため
に設計されたマシンは、そのようなデータ主導ディスパッチを効率的に実⾏するための
dispatch-on-type演算を持っているでしょう。

(test (op self-evaluating?) (reg exp))
(branch (label ev-self-eval))
(test (op variable?) (reg exp))
(branch (label ev-variable))
(test (op quoted?) (reg exp))
(branch (label ev-quoted))
(test (op assignment?) (reg exp))
(branch (label ev-assignment))
(test (op definition?) (reg exp))
(branch (label ev-definition))
(test (op if?) (reg exp))
(branch (label ev-if))
(test (op lambda?) (reg exp))
(branch (label ev-lambda))
(test (op begin?) (reg exp))
(branch (label ev-begin))
(test (op application?) (reg exp))
(branch (label ev-application))
(goto (label unknown-expression-type))

単純な式の評価
(⾃⼰評価の)数値と⽂字列、変数、クォート式、lambda式には、評価するべき
部分式はありません。これらに対しては、評価器は単純に正しい値を valレジ
スタに⼊れ、continueによって指定されたエントリポイントから実⾏を継続
します。単純な式の評価は以下のコントローラコードによって実⾏されます。
ev-self-eval

(assign val (reg exp))
(goto (reg continue))

ev-variable
(assign val (op lookup-variable-value) (reg exp) (reg env))
(goto (reg continue))

ev-quoted
(assign val (op text-of-quotation) (reg exp))
(goto (reg continue))

ev-lambda
(assign unev (op lambda-parameters) (reg exp))
(assign exp (op lambda-body) (reg exp))
(assign val (op make-procedure) (reg unev) (reg exp) (reg env))
(goto (reg continue))

ここで、ev-lambdaは unevと expレジスタを使ってラムダ式の引数と本体を保
持し、env中の環境とともに make-procedure演算に渡せるようにしています。

⼿続き適⽤の評価
⼿続き適⽤は、演算⼦と被演算⼦を持つ組み合わせによって規定されます。演
算⼦は、その値が⼿続きとなる部分式で、被演算⼦は、その値が⼿続き適⽤対
象の引数となる部分式です。メタ循環評価器の evalは、⾃分⾃⾝を再帰的に呼
んで組み合わせの各要素を評価し、それから結果を applyに渡します。apply
が実際の⼿続き適⽤を実⾏します。明⽰制御評価器も同じことを⾏います。再
帰呼び出しは、goto命令と、再帰呼び出しから戻ったときに復元できるように
スタックを使ってレジスタを保存することによって実装します。それぞれの呼
び出しの前には、どのレジスタを保存するべきか注意深く⾒極める必要があり
ます (値が後で必要になるからです)。21

適⽤の評価では、まず演算⼦を評価して⼿続きを⽣成します。これがのち
に、評価済みの被演算⼦に適⽤されることになります。演算⼦を評価するには、
演算⼦を expレジスタに移し、eval-dispatchに進みます。envレジスタに⼊
っている環境は、そのままでも演算⼦を評価する適切な環境になっていますが、
後で被演算⼦の評価にもこの環境が必要になるので、envを保存しておきます。
また、被演算⼦を unevに展開し、それをスタックに保存します。continueは、
演算⼦の評価が終わった後に eval-dispatch が ev-appl-did-operator から
再開できるように設定します。ですがその前に、最初の部分で、適⽤処理が終

21この点は、Lispのような⼿続き型⾔語からレジスタマシン⾔語にアルゴリズムを変
換するにあたって重要なところであり、また難しいところでもあります。必要なものだ
けを保存するというやり⽅以外に、再帰呼び出しの前に (val以外の)すべてのレジスタ
を保存するというやり⽅も考えられます。これは、スタックフレーム (framed-stack)と
いう呼ばれる⽅式です。この⽅式は問題なく動作しますが、必要以上のレジスタを保存
することにもなります。スタック演算が⾼価なシステムでは、これは重要な考慮点にな
りえます。後で中⾝を使う必要のないレジスタを保存すると、本来ガベージコレクショ
ンして領域を解放・再利⽤できたはずの不要なデータを持ち続けることにもつながりま
す。

わった後に継続するべき場所をコントローラに伝える continueの古い値を保
存しておきます。
ev-application

(save continue)
(save env)
(assign unev (op operands) (reg exp))
(save unev)
(assign exp (op operator) (reg exp))
(assign continue (label ev-appl-did-operator))
(goto (label eval-dispatch))

演算⼦の部分式の評価から戻ると、組み合わせの被演算⼦を評価し、返される
引数を arglに保存されるリストに集積する処理に進みます。まず、評価前の
被演算⼦と環境を復元します。arglは空リストに初期化します。それから、演
算⼦を評価することによって⽣成された⼿続きを procレジスタに代⼊します。
被演算⼦がない場合は、直接 apply-dispatchに進みます。被演算⼦がある場
合は、procをスタックに保存し、引数評価ループを開始します。22

ev-appl-did-operator
(restore unev) ; the operands
(restore env)
(assign argl (op empty-arglist))
(assign proc (reg val)) ; the operator
(test (op no-operands?) (reg unev))
(branch (label apply-dispatch))
(save proc)

224.1.3節の評価器データ構造の⼿続きのほかに、引数リストを⾛査する次の⼆つの⼿
続きを追加します。
(define (empty-arglist) '())
(define (adjoin-arg arg arglist) (append arglist (list arg)))

さらに、組み合わせの最後の被演算⼦かどうかのテストを⾏う追加の構⽂⼿続きも使
います。
(define (last-operand? ops) (null? (cdr ops)))

引数評価ループのそれぞれのサイクルでは、unev内のリストから被演算⼦をひ
とつ取得してそれを評価し、その結果を arglに集積していきます。被演算⼦
を評価するときは、被演算⼦を expレジスタに⼊れ、引数集積フェーズから実
⾏を再開できるように continueを設定してから eval-dispatchに進みます。
しかし、まずはそれまでに集積した引数 (argl)、環境 (env)、未評価の残りの
被演算⼦ (unev)を保存します。最後の被演算⼦の評価は特別扱いとなり、ev-
appl-last-argで処理します。
ev-appl-operand-loop

(save argl)
(assign exp (op first-operand) (reg unev))
(test (op last-operand?) (reg unev))
(branch (label ev-appl-last-arg))
(save env)
(save unev)
(assign continue (label ev-appl-accumulate-arg))
(goto (label eval-dispatch))

被演算⼦が評価されると、値が argl内のリストに集積されます。次に、その
被演算⼦は unev内の未評価被演算⼦のリストから削除され、引数評価が続⾏
されます。
ev-appl-accumulate-arg

(restore unev)
(restore env)
(restore argl)
(assign argl (op adjoin-arg) (reg val) (reg argl))
(assign unev (op rest-operands) (reg unev))
(goto (label ev-appl-operand-loop))

最後の引数の評価は異なる処理になります。環境や未評価被演算⼦リストは、
最後の引数を評価した後には必要がなくなるため、eval-dispatchに進む前に
保存する必要はありません。そのため、評価が終わると、ev-appl-accum-last-
argという特別なエントリポイントに帰ります。この⼿続きは引数リストを復
元し、新しい引数を集積し、保存した⼿続きを復元し、適⽤実⾏処理に移動し
ます。23

23最後の被演算⼦の処理を特別扱いするこの最適化は、エブリス末尾再帰 (evlis tail
recursion)として知られています (Wand 1980参照)。最初の被演算⼦も特別扱いすると、

ev-appl-last-arg
(assign continue (label ev-appl-accum-last-arg))
(goto (label eval-dispatch))

ev-appl-accum-last-arg
(restore argl)
(assign argl (op adjoin-arg) (reg val) (reg argl))
(restore proc)
(goto (label apply-dispatch))

組み合わせの中の被演算⼦をインタプリタが評価する順番 (左から右、あるい
は右から左など。練習問題 3.8参照)は、引数評価ループの実装によって決定
されます。メタ循環評価器では、実装に使った基層の Schemeから制御構造を
継承しているため、評価順はメタ循環評価器によって決定されるわけではあ
りません。24 (ev-appl-operand-loopで⼀続きの被演算⼦を unevから抽出す
るために使う) first-operandセレクタは carとして実装されていて、rest-
operandsは cdrとして実装されているので、明⽰制御評価器は組み合わせの
被演算⼦を左から右という順番で評価します。

⼿続き適⽤
エントリポイント apply-dispatchは、メタ循環評価器の apply⼿続きに対応
するものです。apply-dispatchに着くときには、procレジスタには適⽤する
⼿続きが⼊っていて、arglにはその⼿続きの適⽤対象となる評価済み引数の
リストが⼊っています。保存された continueの値 (元は eval-dispatchに渡
され、それが ev-applicationで保存されたもの)がスタック上にあり、これが
⼿続き適⽤の結果を持って帰る先の場所を⽰しています。適⽤が終わると、コ
ントローラは保存された continueの指すエントリポイントに移動します。こ
のとき、valには適⽤の結果が⼊っています。メタ循環評価器の applyと同じ
ように、⼆つの場合を考える必要があります。適⽤する⼿続きが基本⼿続きで
ある場合と、複合⼿続きである場合です。
apply-dispatch

引数評価ループをもう少し効率よくできます。その場合、arglの初期化を最初の被演算
⼦の評価の後まで延期でき、arglを保存することを避けられます。5.5節のコンパイラは
この最適化を実⾏します (5.5.3節の construct-arglist⼿続きと⽐較してください)。

24メタ循環評価器では、被演算⼦の評価順は、4.1.1節の⼿続き list-of-values内の
consに対する引数の評価順によって決定されます (練習問題 4.1参照)。

(test (op primitive-procedure?) (reg proc))
(branch (label primitive-apply))
(test (op compound-procedure?) (reg proc))
(branch (label compound-apply))
(goto (label unknown-procedure-type))

基本⼿続きはそれぞれ、引数を arglから取得し、結果を valに⼊れるよう実
装されていると想定します。マシンの基本⼿続きの扱いを規定するには、本
来ならそれぞれの基本⼿続きを実装するコントローラ命令列を⽤意しておき、
procが指定する基本⼿続きに対応する命令列に、primitive-applyが制御を
移すようにしておく必要があります。しかし、ここで⾒たいのは評価処理の構
造であって、基本⼿続きの細かい実装ではないため、その代わりに、単純に
proc内の⼿続きを argl内の引数に適⽤する apply-primitive-procedure演
算を使うことにします。5.2節のシミュレータを使って評価器のシミュレーシ
ョンを⾏うという⽬的のために、基層の Schemeシステムを呼び出し適⽤を実
⾏する apply-primitive-procedureを使うのですが、これは4.1.4節のメタ循
環評価器で⾏ったことと同じです。基本⼿続きを適⽤して結果が得られたら、
continueを復元して、指定されたエントリポイントに移ります。
primitive-apply

(assign val (op apply-primitive-procedure)
(reg proc)
(reg argl))

(restore continue)
(goto (reg continue))

複合⼿続きの適⽤はメタ循環評価器と同じような進め⽅になります。⼿続きの
仮引数を引数に束縛するフレームを構築し、⼿続きの持つ環境をそのフレーム
によって拡張し、その拡張した環境の中で⼿続きの本体となる式の列を評価し
ます。列の評価は、この先5.4.2節で説明する ev-sequenceで処理します。
compound-apply

(assign unev (op procedure-parameters) (reg proc))
(assign env (op procedure-environment) (reg proc))
(assign env (op extend-environment)

(reg unev) (reg argl) (reg env))
(assign unev (op procedure-body) (reg proc))
(goto (label ev-sequence))

compound-applyは、envレジスタに新しい値が代⼊される、インタプリタ内で
唯⼀の場所です。メタ循環評価器の場合と同じように、新しい環境は、⼿続き
の持っている環境と、引数リストと、それに対応する束縛する変数のリストか
ら構築します。

5.4.2 列の評価と末尾再帰
明⽰制御評価器の ev-sequenceの部分は、メタ循環評価器の eval-sequence
⼿続きと構造が似ています。この⼿続きが扱うのは、⼿続き本体の式の列と、
明⽰的に begin式によって表される式の列です。
明⽰的な begin式は、unevに評価する式の列を⼊れ、continueをスタッ

クに保存し、ev-sequenceにジャンプすることによって評価します。
ev-begin

(assign unev (op begin-actions) (reg exp))
(save continue)
(goto (label ev-sequence))

⼿続き本体の暗黙的な列は、compound-applyから ev-sequenceにジャンプす
ることによって処理します。このとき、continueはすでに ev-applicationで
保存されたものがスタック上にあります。

ev-sequenceと ev-sequence-continueのエントリポイントは、順番に列
のそれぞれの式を評価するループになっています。未評価の式のリストは unev
に保持されています。このリストから⼀つずつ式を取り出して評価するのです
が、評価の前に、評価すべき式が unevにまだ残っているかどうかを確認しま
す。まだあるなら、残りの未評価の式 (unev)と、それらを評価する環境 (env)
を保存してから、eval-dispatchを呼んで式を評価します。この評価から戻る
と、ev-sequence-continueで保存した⼆つのレジスタが復元されます。
列の最後の式は特別扱いとなり、ev-sequence-last-expで処理します。そ

の式の後にはもう評価するべき式はないため、eval-dispatchに⾏く前に unev
と envを保存する必要はありません。最後の式の値が列全体の値となるため、
最後の式の評価が終わった後にするべきことは、現時点でスタックにあるエン
トリポイント (これは ev-application または ev-beginで保存されたもので
す)から継続することだけです。continueを再設定して eval-dispatchがこ
の場所に帰ってくるようにし、その後エントリポイントから継続するためにス
タックから元の continueの値を復元する、という⼿間をかける代わりに、単
に eval-dispatchに⾏く前にスタックから continueを復元しておきます。こ

うすると、eval-dispatchが式を評価した後にすぐにエントリポイントから継
続するようになります。
ev-sequence

(assign exp (op first-exp) (reg unev))
(test (op last-exp?) (reg unev))
(branch (label ev-sequence-last-exp))
(save unev)
(save env)
(assign continue (label ev-sequence-continue))
(goto (label eval-dispatch))

ev-sequence-continue
(restore env)
(restore unev)
(assign unev (op rest-exps) (reg unev))
(goto (label ev-sequence))

ev-sequence-last-exp
(restore continue)
(goto (label eval-dispatch))

末尾再帰
第 1章で、次のような⼿続きによって記述されるようなプロセスは反復プロセ
スであると書きました。
(define (sqrt-iter guess x)

(if (good-enough? guess x)
guess
(sqrt-iter (improve guess x) x)))

この⼿続きは構⽂的には再帰的 (⾃分⾃⾝を使って定義されている)なのです
が、ひとつの sqrt-iter 呼び出しから次の呼び出しに移る際に評価器が情報
を保存する論理的必要性はありません。25 sqrt-iterのような⼿続きが、記
憶領域を増やすことなく⾃分⾃⾝を呼び出し続けられる評価器は、末尾再帰
(tail-recursive)評価器と呼ばれます。第 4章のメタ循環評価器の実装は、評価

255.1節では、そのようなプロセスをスタックを持たないレジスタマシンで実装する⽅
法を学びました。この場合、プロセスの状態は固定のレジスタ集合に格納されます。

器が末尾再帰であるかどうかを規定しませんでした。メタ循環評価器は状態を
保存するメカニズムを基層の Schemeから受け継いでいたからです。しかし、
明⽰制御評価器の場合は、評価プロセスを追うことで、どういった場合に⼿続
き呼び出しがスタック上に情報を累積するかがわかります。
私たちの評価器は末尾再帰です。列の最後の式を評価する際に、スタック

に何も情報を保存せずに直接 eval-dispatchに移動するからです。そのため、
列の最後の式を評価しても—それが⼿続き呼び出しであっても (例えば、sqrt-
iterでは⼿続き本体の最後の式である if式が sqrt-iterの呼び出しに簡約さ
れます)—スタックに何か情報が積まれることはありません。26

この場合に情報を保存する必要はないという事実を利⽤することを考えて
いなければ、eval-sequenceの実装で、列内の式をすべて同じ扱いにしていた
かもしれません。つまり、レジスタを保存し、式を評価し、戻ってレジスタを
復元するということをすべての式の評価が終わるまで繰り返すということで
す。27

ev-sequence
(test (op no-more-exps?) (reg unev))
(branch (label ev-sequence-end))
(assign exp (op first-exp) (reg unev))
(save unev)
(save env)
(assign continue (label ev-sequence-continue))
(goto (label eval-dispatch))

ev-sequence-continue
(restore env)
(restore unev)
(assign unev (op rest-exps) (reg unev))
(goto (label ev-sequence))

26ev-sequenceのこの末尾再帰の実装は、多くのコンパイラで使われる有名な最適化
テクニックの⼀種です。⼿続き呼び出しで終わる⼿続きをコンパイルする際には、その
呼び出しを呼び出される⼿続きのエントリポイントへのジャンプによって置き換えるこ
とができます。この節でやったように、この戦略をインタプリタに組み⼊れると、⾔語
全体で⼀律に最適化が有効になります。

27no-more-exps?は以下のように定義できます。
(define (no-more-exps? seq) (null? seq))

ev-sequence-end
(restore continue)
(goto (reg continue))

これは以前の列評価のコードに対するちょっとした変更のように⾒えるかもし
れません。違うのは、列の最後の式で、ほかの式の場合と同じように、保存・
復元サイクルを通るという点だけです。しかし、これは末尾再帰の実装に対し
ては致命的な変更となります。こうしてしまうと、列の最後の式を評価した後
に、(不要な)保存レジスタを復元するために戻らなければならなくなるからで
す。こうやって余分に保存する分は、⼿続き呼び出しのネストの間に蓄積され
ていきます。結果として、sqrt-iterのようなプロセスが必要とする領域は⼀
定ではなくなり、繰り返し回数に⽐例するようになってしまいます。この違い
が⼤きな影響を及ぼすこともあります。例えば、末尾再帰があれば、無限ルー
プは⼿続き呼び出しのメカニズムだけで表現できます。
(define (count n)

(newline) (display n) (count (+ n 1)))

末尾再帰がなければ、このような⼿続きは最終的にはスタック領域を使い尽く
してしまうことになります。真の反復を表現しようと思うと、⼿続き呼び出し
以外の何らかの制御メカニズムが必要になるでしょう。

5.4.3 条件⽂、代⼊、定義
メタ循環評価器と同じように、特殊形式は式の⼀部を選択的に評価することに
よって処理します。if式の場合は、まず述語を評価して、その値をもとに結果
部と代替部のどちらを評価するか決定します。
述語を評価する前に、後で結果部または代替部を抽出するできるように if

式⾃⾝を保存しておきます。さらに、環境も保存します。これは、結果部また
は代替部を評価する際に必要となります。それに加えて、continueも保存しま
す。これは、ifの値を受け取る式の評価に戻るために後で必要となります。
ev-if

(save exp) ; 後で使うために式を保存
(save env)
(save continue)
(assign continue (label ev-if-decide))
(assign exp (op if-predicate) (reg exp))

(goto (label eval-dispatch)) ; 述語を評価
述語の評価から戻るとき、述語が真であったか偽であったかをテストし、eval-
dispatchに⾶ぶ前にその結果によって結果部か代替部を expに⼊れます。こ
こで envと continueを復元することによって、eval-dispatchが正しい環境
を持つようにし、また if式の値を受け取る正しい場所から継続するようにし
ていることに注意してください。
ev-if-decide

(restore continue)
(restore env)
(restore exp)
(test (op true?) (reg val))
(branch (label ev-if-consequent))

ev-if-alternative
(assign exp (op if-alternative) (reg exp))
(goto (label eval-dispatch))

ev-if-consequent
(assign exp (op if-consequent) (reg exp))
(goto (label eval-dispatch))

代⼊と定義
代⼊と定義は ev-assignment で処理します。ここに到達するのは、eval-
dispatch で exp に代⼊式が⼊っているときです。ev-assignment のコード
は、まず式の値部分を評価し、それから新しい値を環境に組み込みます。set-
variable-value!はマシン演算として利⽤可能であると仮定します。
ev-assignment

(assign unev (op assignment-variable) (reg exp))
(save unev) ; 後で使うために変数を保存
(assign exp (op assignment-value) (reg exp))
(save env)
(save continue)
(assign continue (label ev-assignment-1))
(goto (label eval-dispatch)) ; 代⼊値を評価

ev-assignment-1
(restore continue)

(restore env)
(restore unev)
(perform
(op set-variable-value!) (reg unev) (reg val) (reg env))

(assign val (const ok))
(goto (reg continue))

定義も似たような⽅法で処理します。
ev-definition

(assign unev (op definition-variable) (reg exp))
(save unev) ; 後で使うために変数を保存
(assign exp (op definition-value) (reg exp))
(save env)
(save continue)
(assign continue (label ev-definition-1))
(goto (label eval-dispatch)) ; 定義値を評価

ev-definition-1
(restore continue)
(restore env)
(restore unev)
(perform
(op define-variable!) (reg unev) (reg val) (reg env))

(assign val (const ok))
(goto (reg continue))

練習問題 5.23: 評価器を拡張して、condや letなどの派⽣式を扱
えるようにせよ (4.1.2節)。“ずる”をして、cond->ifのような構⽂
変換器がマシン演算として利⽤可能であると仮定してもよい。28

練習問題 5.24: condを ifに簡約せずに新しい基本特殊形式とし
て実装せよ。真である述語を⾒つけるまで連続した cond節の述語
をテストしていくループを構築し、それから ev-sequenceを使っ
てその節のアクションを評価する必要があるだろう。

28これは、実際にはずるではありません。実際に評価器を⼀から実装するとしても、実
⾏の前の構⽂フェーズで、明⽰制御評価器を使って cond->ifのようなソースレベルの変
換を⾏う Schemeプログラムを解釈させることになるでしょう。

練習問題 5.25: 4.2節の遅延評価器をもとに評価器を修正して、正
規順序評価を使うようにせよ。

5.4.4 評価器の実⾏
明⽰制御評価器の実装で、第 1章から続いてきた開発は終わりを迎えることに
なります。この開発を通して、私たちは評価プロセスのモデルを精緻化してい
くことを⽬指してきました。最初はそれほど厳密でない置換モデルからスター
トし、それから第 3章でこれを環境モデルに拡張しました。環境モデルでは状
態と変更が扱えるようになりました。第 4章のメタ循環評価器では、式の評価
の間に構築される環境構造をより明確にするための⾔語として Schemeそのも
のを使いました。この章では、レジスタマシンを導⼊することで、評価器のメ
モリ管理、引数渡し、制御のメカニズムを詳しく⾒てきました。新しいレベル
の記述に取り組むたびに、それ以前の、そこまで正確ではない評価処理では明
らかになっていなかった問題を提起したり、曖昧性を解決したりする必要があ
りました。明⽰制御評価器のふるまいを理解するにあたっては、それをシミュ
レートしてそのパフォーマンスを監視するということが可能です。
ここでは、この評価器にドライバループを組み込みます。これは、4.1.4節の

driver-loop⼿続きにあたる役割を果たします。評価器はプロンプトを表⽰し、
式を読み込み、eval-dispatchに⾏って式を評価し、結果を表⽰するというこ
とを繰り返します。以下に⽰す命令列が、明⽰制御評価器のコントローラ命令
列の先頭部分になります。29

read-eval-print-loop
(perform (op initialize-stack))
(perform
(op prompt-for-input) (const ";;; EC-Eval input:"))

29ここでは、readやさまざまな表⽰命令が基本マシン演算として利⽤可能であると仮
定しています。これはシミュレーションには便利ですが、実際にはまったく⾮現実的で
す。これらは本当は⾮常に複雑な演算です。実際には、これらの演算は、デバイスとの
間で個別の⽂字をやりとりするような低レベルの⼊出⼒演算を使って実装することにな
るでしょう。
get-global-environment演算をサポートするには、以下のように定義します。

(define the-global-environment (setup-environment))
(define (get-global-environment) the-global-environment)

(assign exp (op read))
(assign env (op get-global-environment))
(assign continue (label print-result))
(goto (label eval-dispatch))

print-result
(perform (op announce-output) (const ";;; EC-Eval value:"))
(perform (op user-print) (reg val))
(goto (label read-eval-print-loop))

⼿続きの中で (apply-dispatchの⽰す “未知の⼿続き型エラー”のような)エ
ラーにぶつかると、エラーメッセージを表⽰してドライバループに戻ります。
30

unknown-expression-type
(assign val (const unknown-expression-type-error))
(goto (label signal-error))

unknown-procedure-type
(restore continue) ; (apply-dispatchの)スタックをクリアする
(assign val (const unknown-procedure-type-error))
(goto (label signal-error))

signal-error
(perform (op user-print) (reg val))
(goto (label read-eval-print-loop))

シミュレーションのために、ドライバループを通るたびにスタックを初期化し
ます。(未定義変数のような)エラーによって評価が中断された後は、スタック
が空でない可能性があるためです。31

5.4.1 節から5.4.4 節までで提⽰したコード⽚をすべてくっつけると、5.2
節のレジスタマシンシミュレータを使って実⾏できる評価器マシンのモデルを
作ることができます。
(define eceval

(make-machine

30インタプリタに扱わせたいエラーはほかにもありますが、それらはそれほど単純で
はありません。練習問題 5.30参照。

31エラーが起こったときだけスタックの初期化を⾏うことも可能です。しかし、ここか
ら先で⾒ていくように、ドライバループの中で初期化するようにすると評価器のパフォ
ーマンスを監視するのに便利です。

'(exp env val proc argl continue unev)
eceval-operations
'(read-eval-print-loop
⟨ここまでで提⽰したマシンコントローラ全体⟩ )))

このほか、評価器が使う基本演算をシミュレートするために Scheme⼿続きを
定義する必要があります。これらは4.1節のメタ循環評価器で使ったのと同じ
⼿続きです。それに加えて、5.4節を通して脚注で定義してきたいくつかの⼿
続きがあります。
(define eceval-operations

(list (list 'self-evaluating? self-evaluating)
⟨明⽰制御評価器の全演算リスト⟩))

これで、グローバル環境を初期化して評価器を実⾏できるようになります。
(define the-global-environment (setup-environment))
(start eceval)
;;; EC-Eval input:
(define (append x y)

(if (null? x) y (cons (car x) (append (cdr x) y))))
;;; EC-Eval value:
ok
;;; EC-Eval input:
(append '(a b c) '(d e f))
;;; EC-Eval value:
(a b c d e f)

もちろん、この⽅法で式を評価すると、Schemeに直接⼊⼒する場合よりもず
っと時間がかかります。この⽅式では何段ものシミュレーションがあるからで
す。式は明⽰制御評価器マシンに評価されるのですが、それは Schemeプログ
ラムによってシミュレートされたもので、そのシミュレートプログラムはそれ
⾃⾝ Schemeインタプリタに評価されることになります。

評価器のパフォーマンス監視
シミュレーションは評価器の実装を導いてくれる強⼒なツールとして使えま
す。また、シミュレーションはレジスタマシンの設計をいろいろ試すのに便利
なだけでなく、シミュレーション対象の評価器のパフォーマンスを監視するの

にも使えます。例えば、パフォーマンスの重要な要因のひとつとして、評価器
がスタックをどれだけ効率的に使えるかということがあります。さまざまな式
を評価するうえで必要となるスタック演算の数は、スタック使⽤の統計を取る
バージョンのシミュレータ (5.2.4節)によって評価器のレジスタマシンを定義
して、さらに評価器の print-result エントリポイントにその統計情報を表⽰
する命令を追加することによって観察できます。
print-result

(perform (op print-stack-statistics)) ; 追加された命令
(perform
(op announce-output) (const ";;; EC-Eval value:"))
. . . ; 以前と同じ

こうすると、評価器とのやりとりは以下のようになります。
;;; EC-Eval input:
(define (factorial n)

(if (= n 1) 1 (* (factorial (- n 1)) n)))
(total-pushes = 3 maximum-depth = 3)
;;; EC-Eval value:
ok
;;; EC-Eval input:
(factorial 5)
(total-pushes = 144 maximum-depth = 28)
;;; EC-Eval value:
120

評価器のドライバループはやりとりのたびにスタックを再初期化するため、表
⽰される統計情報は直前の式の評価に使われたスタック演算のみについてのも
のになります。

練習問題 5.26: 監視機能つきスタックを使って評価器の末尾再帰
の特性 (5.4.2節)を調査せよ。評価器をスタートさせ、1.2.1節の
反復 factorial⼿続きを定義せよ。
(define (factorial n)

(define (iter product counter)
(if (> counter n)

product

(iter (* counter product) (+ counter 1))))
(iter 1 1))

いくつかの⼩さな nの値を使ってこの⼿続きを実⾏せよ。それら
の値の n!を求めるのに必要となった最⼤スタック深度とプッシュ
回数を記録せよ。

a. n!を評価するのに必要な最⼤深度が nから独⽴しているこ
とがわかるだろう。その深さはいくつか。

b. 記録したデータから、任意の n ≥ 1について n!を評価する
のに使われる push演算の総数を表す nの⽅程式を決定せよ。
使われる演算の数は nの線形関数であるため、定数⼆つによ
って決定される。

練習問題 5.27: 練習問題 5.26との⽐較のために、階乗を再帰的に
求める以下の⼿続きのふるまいを調査せよ。
(define (factorial n)

(if (= n 1) 1 (* (factorial (- n 1)) n)))

この⼿続きを監視機能つきのスタックを使って実⾏し、任意の
n ≥ 1に対して n!を評価するのに使われるスタックの最⼤深度と
プッシュの総数を nの関数として決定せよ (この⼆つの関数はま
た線形になる)。以下の表に適切な nの式を埋めて実験結果をまと
めよ。

最⼤深度 pushの総数
階乗

(再帰)
階乗

(反復)
最⼤深度は、計算の実⾏の際に評価器が使う記憶領域の量を表し
ている。プッシュの総数は必要な時間とよく相関している。
練習問題 5.28: eval-sequenceを5.4.2節で記述したように変更し
て評価器の定義を修正し、評価器が末尾再帰でなくなるようにせ
よ。練習問題 5.26と練習問題 5.27の実験を再実⾏し、factorial
⼿続きのどちらのバージョンも、⼊⼒に対して線形に増加する記
憶領域を必要とするようになっていることを⽰せ。

練習問題 5.29: ⽊の再帰によるフィボナッチ数計算のスタック演
算を監視せよ。
(define (fib n)

(if (< n 2)
n
(+ (fib (- n 1)) (fib (- n 2)))))

a. n ≥ 2について Fib(n)を計算するのに必要なスタックの最⼤
深度を nの⽅程式で表せ。ヒント：1.2.2節では、このプロ
セスの使⽤する記憶領域は nに対して線形に増加することを
論じた。

b. n ≥ 2について Fib(n)を計算するのに使われるプッシュの
総数を⽅程式で表せ。プッシュの総数 (これは消費時間とよ
く相関する) が n に対して指数的に増加することがわかる
だろう。ヒント：S(n)を Fib(n)を計算するのに使われるプ
ッシュの総数とする。すると、S(n − 1), S(n − 2), それに
加えて nと独⽴した定数 k という固定の “オーバーヘッド”
を使って S(n)を表す⽅程式があると主張できるだろう。こ
の⽅程式を⽰し、また k の値を求めよ。それから、S(n)が
a · Fib(n + 1) + b として表せることを⽰し、aと bの値を与
えよ。

練習問題 5.30: 今のところ、評価器は⼆種類のエラー—未知の式
の型と未知の⼿続きの型—だけキャッチして知らせるようになっ
ている。ほかのエラーがあると評価器の REPLから抜けてしまう。
この評価器をレジスタマシンシミュレータを使って実⾏している
場合は、これらのエラーは基層の Schemeシステムにキャッチさ
れる。これは、ユーザープログラムがエラーを起こしたときにコ
ンピュータがクラッシュするというのに相当する。32 実際のエラ
ーシステムを動かすというのは⼀⼤プロジェクトになるが、エラ
ーシステムにはどういったことが必要なのかを理解するというこ
とは、その労⼒に⼗分⾒合うものだ。

32残念ながら、C⾔語のような従来のコンパイラベースの⾔語システムではこれが普
通の状況です。unix(tm)ではシステムは “コアをダンプ”し、dos/Windows(tm)ではフ
リーズします。Macintosh(tm)は、爆発する爆弾の絵を表⽰して、コンピュータを再起
動する機会を与えてくれます —運がよければですが。

a. 評価プロセスで起こるエラー (未束縛変数へのアクセスの試
⾏など)は、探索演算に修正を加えて、特別な状態コードを
返すようにすることでキャッチできる。状態コードはユーザ
ー変数の取りうる値であってはならない。評価器はこの状態
コードをテストし、signal-errorに⾏くために必要なこと
を⾏う。評価器の中でそのような変更が必要となる箇所をす
べて探して修正せよ。これは⼤変な作業だ。

b. もっと⼤変なのは、基本⼿続き適⽤によって発せられるエ
ラーの扱いという問題だ。そのようなエラーには、ゼロ除算
の試⾏や、記号から car を取り出そうとすることなどがあ
る。プロによって書かれた⾼品質システムでは、基本⼿続き
の適⽤は、その基本⼿続きの⼀部として、安全チェックが⾏
われている。例えば、carの呼び出しでは、引数がペアであ
るかをまずチェックするようにできるだろう。引数がペアで
なければ、その適⽤は評価器に特別な状態コードを返し、そ
れを受けて評価器が失敗を報告する。レジスタマシンシミュ
レータでは、それぞれの基本⼿続きが適⽤可能性チェックを
⾏い、失敗時には特別な状態コードを適切に返すようにす
ることによって、これが実現できる。こうすると、評価器の
primitive-applyコードは、状態コードをチェックしたうえ
で必要であれば signal-errorに⾏くようにできる。この構
造を構築し、動くようにせよ。これは⼀⼤プロジェクトだ。

5.5 コンパイル
5.4節の明⽰制御評価器は、コントローラが Schemeプログラムを解釈するレ
ジスタマシンです。この節では、コントローラが Schemeインタプリタではな
いレジスタマシンで Schemeプログラムを動かす⽅法について⾒ていきます。
明⽰制御評価器は万能です。この評価器は、Schemeで記述可能な計算プロ

セスであればどんなものでも実⾏できます。評価器のコントローラは、望み通
りの計算を実⾏するために、評価器のデータパスの使い⽅を指揮します。この
ため、評価器のデータパスは万能です。適切なコントローラがあれば、望み通
りの計算を⼗分実⾏できるのです。33

33これは理論的にはそうだということです。この評価器のデータパスが汎⽤コンピュ

商⽤の汎⽤コンピュータは、効率的で便利な万能データパスを構成するレ
ジスタと演算の集合を中⼼として構成されているレジスタマシンです。汎⽤マ
シンのコントローラは、ここまで使ってきたようなレジスタマシン⾔語のイン
タプリタです。この⾔語はマシンのネイティブ⾔語 (native language)、または
単純に機械語 (machine language)と呼ばれます。機械語で記述されたプログ
ラムは、そのマシンのデータパスを使う命令の列です。例えば、明⽰制御評価
器の命令列は、特化したインタプリタマシンのコントローラとして考えるので
はなく、汎⽤⽬的のコンピュータの機械語プログラムとして考えることができ
ます。
⾼級⾔語とレジスタマシン⾔語の橋渡しをする戦略として⼀般的なものは

⼆つあります。明⽰制御評価器が⽰しているのは解釈という戦略です。機械の
ネイティブ⾔語で書かれたインタプリタ (解釈器) は、評価を実⾏するマシン
のネイティブ⾔語とは異なる⾔語 (ソース⾔語 (source language))で書かれた
プログラムを実⾏するようにマシンを設定します。ソース⾔語の基本⼿続きは、
マシンのネイティブ⾔語によって書かれたサブルーチンライブラリとして実装
されます。解釈対象のプログラム (ソースプログラム (source program))はデー
タ構造として表現されます。インタプリタはこのデータ構造をたどりながらソ
ースプログラムの構造を解析していきます。その過程で、インタプリタはライ
ブラリから適切な基本⼿続きのサブルーチンを呼ぶことによってソースプログ
ラムの意図されたふるまいをシミュレートします。
この節で⾒ていくのは、コンパイル (compilation)というもうひとつの戦略

です。あるソース⾔語とマシンに対するコンパイラは、ソースプログラムをそ
のマシンのネイティブ⾔語で書かれた等価なプログラム (オブジェクトプログ
ラム (object program))に変換します。この節で実装するコンパイラは、Scheme
で書かれたプログラムを明⽰制御評価器のデータパスを使って実⾏する命令列
に変換するものです。34

ータのデータパスとして特に便利であるとか効率的であるとか主張しているわけではあ
りません。例えば、このデータパスは⾼性能浮動⼩数点計算やビットベクタを集中的に
⾛査するような計算を実装するのにはあまり向いていません。

34実際は、コンパイルされたコードを実⾏するマシンはインタプリタのマシンよりも
単純にできます。コードを実⾏するマシンでは expレジスタと unevレジスタを使わない
からです。インタプリタはこれらを使って未評価の式の⼀部を保持していました。しか
し、コンパイラの場合はそれらの式はレジスタマシンが実⾏するコンパイル済みコード
の中に組み込まれることになります。それと同じ理由で、式の構⽂を扱う機械語演算も
必要なくなります。しかし、コンパイル済みのコードは、明⽰制御評価器にはなかった
(コンパイル済み⼿続きオブジェクトを表現するための)機械語演算を追加でいくつか使
います。

解釈という戦略に⽐べると、コンパイルという戦略はプログラム実⾏の効
率性を⼤きく向上させることができます。このことはのちにコンパイラの概観
のところで説明します。⼀⽅、インタプリタは対話的にプログラムを開発した
りデバッグしたりするためのずっと強⼒な環境を提供してくれます。実⾏され
ているソースプログラムを実⾏時に検査したり修正したりできるからです。そ
れに加えて、基本⼿続きのライブラリ全体があるため、デバッグ中に新しいプ
ログラムを構築してシステムに追加するということも可能です。
コンパイルと解釈には互いに補う利点があるという観点から、現代のプロ

グラム開発環境は混合戦略を⽬指しています。Lispインタプリタは、⼀般的に
解釈される⼿続きとコンパイル済みの⼿続きが互いに呼び出せるように構成さ
れています。そのおかげで、プログラマはデバッグ済みと想定されるプログラ
ムの部分をコンパイルして効率性という利点を得ると同時に、対話的開発とデ
バッグという流れの中にあるプログラムの部分の実⾏をインタプリタモードの
ままにしておくということができます。コンパイラを実装した後の5.5.7節で
は、コンパイラとインタプリタと接続して統合的なインタプリタ-コンパイラ
開発システムを作る⽅法を⽰します。

コンパイラの概要
ここで作るコンパイラは、構造と実⾏する機能の両⾯で、ここまでで作ったイ
ンタプリタによく似ています。そのため、コンパイラが式を解析するメカニズ
ムはインタプリタのものと似たものになります。さらに、コンパイル済みコー
ドと解釈されるコードとの接続を簡単にするため、出⼒するコードがインタプ
リタのレジスタの⽤法と同じ約束に従うものになるようにコンパイラを設計し
ます。つまり、環境は envレジスタに保持し、引数リストは arglに集積し、適
⽤する⼿続きは procに⼊れ、⼿続きは答えを valに⼊れて戻り、⼿続きが戻
るべき場所は continueに⼊れるようにするということです。全体として、コ
ンパイラがソースプログラムをオブジェクトプログラムに翻訳するとき、イン
タプリタが同じソースプログラムを評価する際に⾏われるのと本質的に同じレ
ジスタ演算をオブジェクトプログラムが⾏うように翻訳を⾏います。
このように考えると、基本的なコンパイラを実装する戦略が⾒えてきます。

まず、インタプリタと同じように式をたどっていきます。そして、インタプリ
タであれば式の評価中に⾏うであろうレジスタ命令にぶつかると、その命令の
実⾏はせずに、その代わりにそれを列に集積していきます。その結果としてで
きる命令列がオブジェクトコードになります。コンパイルが解釈に対して性能
⾯での利点があるということがわかるでしょうか。インタプリタが式—例えば

(f 84 96)—を評価するたびに、式の分類 (これが⼿続き適⽤であるというこ
とを知る)と被演算⼦リストの終端テスト (被演算⼦が⼆個であるということ
を知る)という仕事を⾏います。コンパイラの場合、式はコンパイル時に命令
列が⽣成されるときに⼀度解析されるだけです。コンパイラに⽣成されたオブ
ジェクトコードは、演算⼦と⼆つの被演算⼦を評価し、引数リストを組み⽴て、
(procにある)⼿続きを (arglにある)引数に適⽤するという命令しか含んでい
ません。
これは、4.1.7節の解析評価器で実装したものと同じ種類の最適化です。し

かし、コンパイル済みコードでは、さらに効率を上げられる機会があります。
インタプリタの実⾏時には、⾔語のどんな式に対しても使える必要があるプロ
セスに従って動作しています。それに対して、コンパイル済みコードのある⼀
部は、ある特定の式を実⾏することが想定されています。これは⼤きな違いに
なることもあります。例えば、スタックを使ってレジスタを保存する場合です。
インタプリタが式を評価する際には、インタプリタはどんな事態に対しても備
えておく必要があります。部分式を評価する前には、インタプリタは後で必要
になるレジスタをすべて保存します。これは、部分式がどんな評価を必要とす
るかわからないからです。⼀⽅、コンパイラの場合は、処理している特定の式
の構造を利⽤して、不必要なスタック演算を⾏わないコードを⽣成できます。
⼀例として、組み合わせ (f 84 96)について考えてみましょう。インタプ

リタが組み合わせの演算⼦を評価する前に、被演算⼦と環境を持つレジスタを
保存してこの評価に備えます。これらのレジスタの値は後で必要になります。
次に、インタプリタは演算⼦を評価して、その結果を valに取得し、保存した
レジスタを復元し、最後に結果を valから procに移します。しかし、いま扱
っているこの特定の式では、演算⼦は記号 fであり、その評価は機械語演算の
lookup-variable-valueによって⾏われます。この演算はどのレジスタの値も
変化させません。この節で実装するコンパイラはこのことを利⽤して、次の命
令によって演算⼦を評価するコードを⽣成します。
(assign proc (op lookup-variable-value)

(const f)
(reg env))

このコードは、不必要な保存と復元をしないで済ませるほかに、検索した値を
直接 procに割り当てるということもしています。インタプリタであれば、結
果を valの中に取得し、その後 procに移すところです。
コンパイラは環境へのアクセスも最適化できます。コードを解析した状態

では、コンパイラは多くの場合どのフレームに特定の変数があるかを知ってい

て、lookup-variable-valueによる検索を実⾏せずに、そのフレームに直接ア
クセスできます。そのような変数アクセスの実装⽅法については 5.5.6節で検
討します。しかし、それまでの間は、上で記述したようなレジスタとスタック
の最適化に焦点を合わせることにします。コンパイラが⾏える最適化はほかに
も数多くあります。その中には、applyメカニズム (練習問題 5.38参照)を使わ
ずに、基本演算を “インライン”でコードするといったものがあります。しか
し、ここでは深⼊りしないことにしましょう。この章の主な⽬的は、単純な (そ
れでいて⾯⽩さを失わない)⽂脈でコンパイル処理を説明するというところに
あります。

5.5.1 コンパイラの構造
4.1.7節では、元のメタ循環インタプリタに修正を加えて、分析と実⾏を分離し
ました。それぞれの式を分析し、環境を引数として必要な計算を⾏う実⾏⼿続
きを⽣成するようにしました。ここでのコンパイラも、本質的に同じ解析を⾏
います。しかし、実⾏⼿続きを⽣成するのではなく、レジスタマシンで動かす
命令列を⽣成することになります。
コンパイラのトップレベルのディスパッチは、compileという⼿続きです。

これは、4.1.1節の eval⼿続きや、4.1.7節の analyze⼿続き、5.4.1節の明⽰
制御評価器の eval-dispatchエントリポイントに対応するものです。このコン
パイラは、インタプリタと同じように、4.1.2節で定義した式構⽂⼿続きを使い
ます。35 compileは、コンパイル対象の式の構⽂型について場合分けを⾏い、
それぞれの式の型を、それらに専⾨化したコード⽣成器 (code generator)にデ
ィスパッチします。
(define (compile exp target linkage)

(cond ((self-evaluating? exp)
(compile-self-evaluating exp target linkage))

((quoted? exp) (compile-quoted exp target linkage))
((variable? exp)
(compile-variable exp target linkage))

((assignment? exp)

35このコンパイラは Schemeプログラムであり、式を操作するのに使う構⽂⼿続きは
メタ循環評価器で使った実際の Scheme⼿続きです。それに対して、明⽰制御評価器の
ときには、それらと等価な構⽂演算がレジスタマシンの演算として利⽤可能であると仮
定していました (もちろん、レジスタマシンを Schemeでシミュレートしたときには、レ
ジスタマシンのシミュレーションに実際の Scheme⼿続きを使いました)。

(compile-assignment exp target linkage))
((definition? exp)
(compile-definition exp target linkage))

((if? exp) (compile-if exp target linkage))
((lambda? exp) (compile-lambda exp target linkage))
((begin? exp)
(compile-sequence
(begin-actions exp) target linkage))

((cond? exp)
(compile (cond->if exp) target linkage))

((application? exp)
(compile-application exp target linkage))

(else
(error "Unknown expression type: COMPILE" exp))))

ターゲットとリンク
compileと、compileが呼ぶコード⽣成器は、コンパイル対象の式に加えて⼆
つの引数を取ります。ひとつはターゲット (target) で、コンパイル済みコー
ドが式の値を返すレジスタを指定します。もうひとつはリンク記述⼦ (linkage
descriptor)で、式をコンパイルした結果のコードが実⾏を終えたときにどのよ
うな動作をするか指定します。リンク記述⼦は、以下の三つのうちのひとつの
動作をするようコードに要求します。

• 列の次の命令から実⾏を続ける (これはリンク記述⼦ nextによって規定
します)

• コンパイルしている⼿続きから戻る (これはリンク記述⼦ returnによっ
て規定します)

• 名前つきエントリポイントに⾶ぶ (これは指定したラベルをリンク記述
⼦として使うことによって規定します)

例えば、ターゲットをレジスタ val、リンク記述⼦を nextとして、(⾃⼰評価
の)式 5をコンパイルすると、以下の命令が⽣成されます。
(assign val (const 5))

同じ式を、リンク記述⼦を returnとしてコンパイルすると、以下の命令が⽣
成されます。

(assign val (const 5))
(goto (reg continue))

⼀つ⽬の場合は、実⾏は列の中の次の命令から続⾏されます。⼆つ⽬の場合は、
⼿続き呼び出しから戻ることになります。いずれの場合も、式の値はターゲッ
トレジスタ valに⼊れられます。

命令列とスタックの使⽤
それぞれのコード⽣成器は、式に対して⽣成したオブジェクトコードを含む命
令列 (instruction sequence)を返します。複合式に対するコード⽣成は、部分式
のより単純なコード⽣成器からの出⼒を組み合わせることによって達成されま
す。これは、複合式の評価が部分式を評価することによって達成されるのと同
じです。
命令列を組み合わせる最も単純な⼿法は、append-instruction-sequences

という⼿続きです。これは順に実⾏する任意の数の命令列を引数として取り、
それらを連結し、組み合わせた列を返します。つまり、⟨seq1⟩と ⟨seq2⟩が命令
列であるとすると、以下を評価すると、
(append-instruction-sequences ⟨seq1⟩ ⟨seq2⟩)

次の列が⽣成されることになります。
⟨seq1⟩
⟨seq2⟩

レジスタを保存する必要があるかもしれないときは、コンパイラのコード⽣成
器は preservingを使います。これは、命令列の組み⽴てを⾏うためのより込
み⼊った⼿法です。preservingは三つの引数を取ります。ひとつのレジスタの
集合と、⼆つの命令列です。この⼿続きは、もし⼆つ⽬の命令列の実⾏に必要
であれば、⼀つ⽬の命令列を実⾏する際に集合内のレジスタの中⾝が保存され
るようなやり⽅で、列を連結します。これはつまり、もし⼀つ⽬の命令列がレ
ジスタを書き換え、⼆つ⽬の命令列が実際にそのレジスタの元の中⾝を必要と
するのであれば、preservingは列を連結する前に最初の列をそのレジスタの
saveと restoreによって囲むということです。そうでなければ、preserving
は単純に連結した命令列を返します。よって、例えば
(preserving (list ⟨reg1⟩ ⟨reg2⟩) ⟨seg1⟩ ⟨seg2⟩)

は、⟨seq1⟩と ⟨seq2⟩が ⟨reg1⟩と ⟨reg2⟩をどのように使うかによって、以下
の四つの命令列のうちのひとつを⽣成します。

⟨seq1⟩ (save ⟨reg1⟩) (save ⟨reg2⟩) (save ⟨reg2⟩)
⟨seq2⟩ ⟨seq1⟩ ⟨seq1⟩ (save ⟨reg1⟩)

(restore ⟨reg1⟩) (restore ⟨reg2⟩) ⟨seq1⟩
⟨seq2⟩ ⟨seq2⟩ (restore ⟨reg1⟩)

(restore ⟨reg2⟩)
⟨seq2⟩

preservingを使って命令列を組み合わせることによって、コンパイラは不必
要なスタック命令を避けることができます。また、こうすることによって save
と restoreという命令を⽣成するかどうかという実装上の詳細を preserving
⼿続きの中に分離し、これらの問題を個別のコード⽣成器を書くときに起こる
考慮点から切り離すことができます。実際に、saveと restoreという命令を
コード⽣成器が明⽰的に⽣成することはありません。
原則として、命令列は単純な命令のリストとして表現できるはずです。そ

うすると、append-instruction-sequencesは通常のリストの appendを実⾏
することによって命令列の組み⽴てができるところです。しかし、その場合、
preserving は複雑な演算になってしまいます。それぞれの命令列を解析し
て、命令列がどのようにレジスタを使うのか調べなければならないからです。
preservingは、複雑なうえに⾮効率的なものになります。というのは、引数の
命令列はそれ⾃⾝ preservingの呼び出しによって構築されたものかもしれず、
その場合はそれらの部品はすでに解析されているはずなのですが、それをまた
解析しなければならないからです。そのような繰り返しの多い解析を避けるた
めに、それぞれの命令列に、レジスタ使⽤情報を関連づけることにします。基
本的な命令列を構築する際にこの情報を明⽰的に提供するようにして、命令列
を組み合わせる⼿続きは、部品となる命令列に関連づけられた情報から組み合
わせた命令列のレジスタ使⽤情報を導き出します。
命令列は三つの情報を持つことになります。
• 命令列内の命令が実⾏される前に初期化しなければならないレジスタ集
合 (これらのレジスタは命令列が 必要とする (needed)と⾔います)

• 列内の命令によって値が変更されるレジスタ集合
• 列内の実際の命令 (命令⽂ (statement)とも⾔います)

命令列は、三つの部品のリストとして表現します。よって、命令列のコンスト
ラクタは以下のようになります。
(define (make-instruction-sequence

needs modifies statements)
(list needs modifies statements))

例えば、現在の環境内で変数 xの値を検索し、その結果を valに割り当て、そ
れから戻るという⼆命令の列は、レジスタ envと continueが初期化されてい
る必要があり、またレジスタ valを変更します。したがって、この列は以下の
ように構築されることになります。
(make-instruction-sequence
'(env continue)
'(val)
'((assign val

(op lookup-variable-value) (const x) (reg env))
(goto (reg continue))))

場合によっては、命令⽂を持たない命令列を構築することが必要になることも
あります。
(define (empty-instruction-sequence)

(make-instruction-sequence '() '() '()))

命令列を組み⽴てる⼿続きは5.5.4節で⽰します。
練習問題 5.31:⼿続き適⽤を評価する際、明⽰制御評価器は常に演
算⼦の評価の前後で envレジスタの保存と復元を⾏い、(最後の 1
つを除いて)各被演算⼦の評価の前後で envの保存と復元を⾏い、
各被演算⼦の評価の前後で arglの保存と復元を⾏い、被演算⼦列
の評価の前後で procの保存と復元を⾏う。以下のそれぞれの組み
合わせに対して、これらの saveと restore命令のうち、余分であ
り、そのためコンパイラの preservingメカニズムによって取り除
くことができるものを答えよ。
(f 'x 'y)
((f) 'x 'y)
(f (g 'x) y)
(f (g 'x) 'y)

練習問題 5.32: preservingメカニズムを使うと、組み合わせの演
算⼦が記号である場合、コンパイラは演算⼦の評価の前後で env
の保存と復元を回避することになる。このような最適化は、評価
器にも組み込むことができるものだ。実際に、5.4節の明⽰制御評
価器は、被演算⼦のない組み合わせを特別扱いすることによって、
すでに似たような最適化を⾏っている。

a. 明⽰制御評価器を拡張し、演算⼦が記号である組み合わせを
独⽴したクラスの式として認識するようにし、そのような式
を評価する際にその事実を利⽤するようにせよ。

b. Alyssa P. Hackerは、評価器が認識する特別ケースをどんど
ん増やしていけば、コンパイラの最適化をすべて組み込むこ
とができ、そうするとコンパイルをすることの利点は完全に
なくなるのではないかという考えを述べた。あなたはこの考
え⽅についてどう思うか。

5.5.2 式のコンパイル
この節と次の節では、compile⼿続きのディスパッチ先となるコード⽣成器を
実装していきます。

リンクコードのコンパイル
⼀般的に、それぞれのコード⽣成器の出⼒は、要求されたリンクを実装する命
令で終わります。これは⼿続き compile-linkageによって⽣成されます。リ
ンクが returnであれば、命令 (goto (reg continue))を⽣成しなければなり
ません。これは continueレジスタを必要とし、どのレジスタも変更しません。
リンク記述⼦が nextであれば、命令を追加する必要はありません。これら以
外の場合、リンク記述⼦はラベル、そのラベルへの gotoを⽣成します。この
命令はどのレジスタも必要とせず、変更もしません。36

36この⼿続きは、バッククォート (backquote)(または擬似クォート (quasiquote))とい
うリスト構築に便利な Lispの機能を使っています。リストの前にバッククォート記号を
置くというのはクォートにとても似ていますが、リストの中でカンマによって印をつけ
られたものは評価されるというところが違います。
例えば、linkageの値が記号 branch25である場合、以下の式を評価すると

`((goto (label ,linkage)))

(define (compile-linkage linkage)
(cond ((eq? linkage 'return)

(make-instruction-sequence '(continue) '()
'((goto (reg continue)))))

((eq? linkage 'next)
(empty-instruction-sequence))

(else
(make-instruction-sequence '() '()
`((goto (label ,linkage)))))))

リンクのコードは、preservingによって continueレジスタを保存したうえで
追加されます。これは、returnリンクが continueレジスタを必要とするから
です。与えられた命令列が continueを変更するものであり、またリンクのコ
ードが continueを必要とする場合、continueは保存と復元が⾏われます。
(define (end-with-linkage linkage instruction-sequence)

(preserving '(continue)
instruction-sequence
(compile-linkage linkage)))

単純な式のコンパイル
⾃⼰評価式、クォート、変数に対するコード⽣成器は、必要な値をターゲット
レジスタに割り当て、リンク記述⼦によって規定された通りに進む命令列を構
築します。
(define (compile-self-evaluating exp target linkage)

(end-with-linkage linkage

次のリストになります。
((goto (label branch25)))

同じように、xの値がリスト (a b c)である場合、以下の式を評価すると
`(1 2 ,(car x))

次のリストになります。
(1 2 a)

(make-instruction-sequence '() (list target)
`((assign ,target (const ,exp))))))

(define (compile-quoted exp target linkage)
(end-with-linkage linkage
(make-instruction-sequence '() (list target)
`((assign ,target (const ,(text-of-quotation exp)))))))

(define (compile-variable exp target linkage)
(end-with-linkage linkage
(make-instruction-sequence '(env) (list target)
`((assign ,target

(op lookup-variable-value)
(const ,exp)
(reg env))))))

これらの代⼊命令はすべてターゲットレジスタを変更します。また、変数の検
索を⾏う命令は env レジスタを必要とします。
代⼊と定義は、インタプリタでの場合とだいたい同じように扱われます。

変数に割り当てる値を求めるコードを再帰的に⽣成し、それに⼆命令からなる
列を追加します。この命令列は、実際に変数の設定や定義を⾏い、式全体の値
(記号 ok)をターゲットレジスタに代⼊します。再帰的コンパイルはターゲット
valとリンク記述⼦ nextを持つので、コードは結果を valに⼊れ、その後に
追加されるコードから続⾏します。追加は envを保存して⾏われます。これは、
変数の設定や定義には環境が必要であり、また変数の値のコードはレジスタを
任意に変更するような複雑な式のコンパイルとなるかもしれないからです。
(define (compile-assignment exp target linkage)

(let ((var (assignment-variable exp))
(get-value-code
(compile (assignment-value exp) 'val 'next)))

(end-with-linkage linkage
(preserving '(env)
get-value-code
(make-instruction-sequence '(env val) (list target)
`((perform (op set-variable-value!)

(const ,var)
(reg val)
(reg env))

(assign ,target (const ok))))))))
(define (compile-definition exp target linkage)

(let ((var (definition-variable exp))
(get-value-code
(compile (definition-value exp) 'val 'next)))

(end-with-linkage linkage
(preserving '(env)
get-value-code
(make-instruction-sequence '(env val) (list target)
`((perform (op define-variable!)

(const ,var)
(reg val)
(reg env))

(assign ,target (const ok))))))))

追加される⼆命令の列は、envと valを必要とし、ターゲットを変更します。
この命令列のために envは保存するのに、valは保存しないということに気を
つけてください。これは、この命令列が使うために、get-value-codeが結果を
明⽰的に valに⼊れるよう設計されていることによります (実際、もし valを
保存するとバグになります。こうしてしまうと、get-value-codeの実⾏直後
に valの以前の中⾝が復元されるようになってしまうからです)。

条件式のコンパイル
与えられたターゲットとリンク記述⼦でコンパイルされた if式のコードは以
下の形式になります。
⟨ターゲット val, リンク記述⼦ next での述語のコンパイル⟩
(test (op false?) (reg val))
(branch (label false-branch))

true-branch
⟨与えられたターゲットと、与えられたリンク記述⼦または after-ifでの結
果部のコンパイル⟩
false-branch
⟨与えられたターゲットとリンク記述⼦での代替部のコンパイル⟩

after-if

このコードを⽣成するには、述語、結果部、代替部をコンパイルし、その結果
できるコードを、述語の結果をテストする命令、真と偽の分岐と条件⽂の終わ
りの印となる新しく⽣成されたラベルと組み合わせます。37 このコードの配置
では、テストが偽であれば、真の分岐を⾶び越す必要があります。ここでひと
つだけ注意するべきところは、真の分岐のリンクをどう扱うかということです。
条件⽂のリンクが return またはラベルの場合、真と偽の分岐は両⽅ともこの
同じリンクを使います。しかし、リンクが nextの場合、真の分岐は、偽の分岐
を⾶び越し条件⽂の最後に⾶ぶコードで終わることになります。
(define (compile-if exp target linkage)

(let ((t-branch (make-label 'true-branch))
(f-branch (make-label 'false-branch))
(after-if (make-label 'after-if)))

(let ((consequent-linkage
(if (eq? linkage 'next) after-if linkage)))

(let ((p-code (compile (if-predicate exp) 'val 'next))
(c-code
(compile
(if-consequent exp) target

consequent-linkage))
(a-code
(compile (if-alternative exp) target linkage)))

(preserving '(env continue)

37ここで、true-branch, false-branch, after-ifというラベルを上で⽰したようにそ
のまま使うことはできません。プログラム中に if⽂は複数あるかもしれないからです。
make-labelは、記号を引数に取り、与えられた記号で始まる新しい記号を返します。例
えば、(make-label 'a)を連続して呼び出すと、a1, a2, . . .を返します。make-labelは、
クエリ⾔語で⼀意の変数名を⽣成したときと同じように、以下のように実装できます。
(define label-counter 0)
(define (new-label-number)

(set! label-counter (+ 1 label-counter))
label-counter)

(define (make-label name)
(string->symbol

(string-append (symbol->string name)
(number->string (new-label-number)))))

p-code
(append-instruction-sequences
(make-instruction-sequence '(val) '()
`((test (op false?) (reg val))

(branch (label ,f-branch))))
(parallel-instruction-sequences
(append-instruction-sequences t-branch c-code)
(append-instruction-sequences f-branch a-code))

after-if))))))

envは、真と偽の分岐で必要とされるかもしれないので、述語コードの前後で
保存されます。continueは、真と偽の分岐に含まれるリンクコードで必要とさ
れるかもしれないので保存されます。(順に実⾏されるわけではない)真と偽の
分岐のコードは、5.5.4節で記述する parallel-instruction-sequencesとい
う専⽤の結合器を使って連結されます。

cond は派⽣式なので、その処理のためにコンパイラがするのは、(4.1.2
節の) cond->if変換器を適⽤し、結果の if式をコンパイルすることだけです。

列のコンパイル
列 (⼿続きの本体または明⽰的な begin式)のコンパイルは、列の評価と同じ
ような構造です。列のそれぞれの式がコンパイルされるとき、最後の式は列
全体に対して指定されたリンクでコンパイルされ、ほかの式は (列の残りを実
⾏するために)next リンクでコンパイルされます。個別の式の命令列は、(列
の残りのために必要な)envと (列末尾のリンクコードで必要となるかもしれな
い)continueを保存しながら、ひとつの命令列に連結されます。
(define (compile-sequence seq target linkage)

(if (last-exp? seq)
(compile (first-exp seq) target linkage)
(preserving
'(env continue)
(compile (first-exp seq) target 'next)
(compile-sequence (rest-exps seq) target linkage))))

lambda式のコンパイル
lambda式は⼿続きを構築します。lambda式のオブジェクトコードは以下の形
式に従う必要があります。
⟨⼿続きオブジェクトを構築しそれをターゲットレジスタに割り当てる⟩
⟨リンク⟩
lambda式をコンパイルするときには、⼿続き本体のコードも⽣成します。本
体は⼿続き構築時に実⾏されるわけではないですが、これをオブジェクトコー
ドの lambda式のコードの直後に挿⼊しておくと便利です。lambda式のリンク
がラベルまたは returnの場合、これは問題になりません。しかし、リンクが
nextの場合、⼿続き本体の後ろに挿⼊されたラベルまでジャンプするリンクを
使って、⼿続き本体のコードを⾶び越す必要があります。よって、オブジェク
トコードは以下の形式になります。
⟨⼿続きオブジェクトを構築しそれをターゲットレジスタに割り当てる⟩
⟨与えられたリンクのコード⟩ または (goto (label after-lambda))
⟨⼿続き本体をコンパイルしたもの⟩

after-lambda

compile-lambdaは、⼿続きオブジェクトに⼿続き本体のコードが続くものを
構築するコードを⽣成します。⼿続きオブジェクトは、現在の環境 (lambda式
定義時の環境)とコンパイル済みの⼿続き本体を指すエントリポイント (新し
く⽣成されたラベル)を引数として、実⾏時に構築されることになります。38

(define (compile-lambda exp target linkage)
(let ((proc-entry (make-label 'entry))

(after-lambda (make-label 'after-lambda)))

38ここで、コンパイル済み⼿続きを表現するデータ構造を実装する機械語演算がいく
つか必要になります。データ構造は、4.1.3節で記述した複合⼿続きの構造と同じような
ものになります。
(define (make-compiled-procedure entry env)

(list 'compiled-procedure entry env))
(define (compiled-procedure? proc)

(tagged-list? proc 'compiled-procedure))
(define (compiled-procedure-entry c-proc) (cadr c-proc))
(define (compiled-procedure-env c-proc) (caddr c-proc))

(let ((lambda-linkage
(if (eq? linkage 'next) after-lambda linkage)))

(append-instruction-sequences
(tack-on-instruction-sequence
(end-with-linkage lambda-linkage
(make-instruction-sequence '(env) (list target)
`((assign ,target

(op make-compiled-procedure)
(label ,proc-entry)
(reg env)))))

(compile-lambda-body exp proc-entry))
after-lambda))))

compile-lambda が⼿続きの本体を lambda 式のコードに連結する際には、
append-instruction-sequences(5.5.4 節) ではなく、特別な結合器 tack-on-
instruction-sequenceを使います。⼿続き本体は、組み⽴てた列に⼊ったと
きに実⾏される命令列の⼀部ではないからです。⼿続き本体が列の中にあるの
は、単純に置く場所として便利だからというだけです。

compile-lambda-bodyは⼿続き本体のコードを構築します。このコードは、
エントリポイントのラベルから始まります。その次に、⼿続き本体を評価する
正しい環境、つまり⼿続きの定義環境に仮引数と⼿続き呼び出しの引数との束
縛を加えて拡張した環境に実⾏時の評価環境を切り替えさせる命令が続きま
す。その次には、⼿続き本体となる式の列のコードが続きます。列は、⼿続き
の結果が valに⼊った状態で最後に⼿続きからのリターンで終わるように、リ
ンク returnとターゲット valでコンパイルされます。
(define (compile-lambda-body exp proc-entry)

(let ((formals (lambda-parameters exp)))
(append-instruction-sequences
(make-instruction-sequence '(env proc argl) '(env)
`(,proc-entry

(assign env
(op compiled-procedure-env)
(reg proc))

(assign env
(op extend-environment)
(const ,formals)

(reg argl)
(reg env))))

(compile-sequence (lambda-body exp) 'val 'return))))

5.5.3 組み合わせのコンパイル
コンパイル処理の核⼼部分は⼿続き適⽤のコンパイルです。与えられたターゲ
ットとリンク記述⼦でコンパイルされた組み合わせのコードは以下の形式を持
ちます。
⟨ターゲット proc, リンク nextで演算⼦をコンパイル⟩
⟨被演算⼦を評価し、arglに引数リストを構築⟩
⟨与えられたターゲットとリンクで⼿続き呼び出しをコンパイル⟩
レジスタ env, proc, arglは、演算⼦と被演算⼦を評価する間に保存と復元を
⾏う必要があるかもしれません。このコンパイラで、val以外のターゲットが
指定される場所はここだけです。
必要なコードは compile-applicationによって⽣成されます。これは、演

算⼦をコンパイルして適⽤するべき⼿続きを proc に⼊れるコードを⽣成し、
被演算⼦をコンパイルして適⽤の個々の被演算⼦を評価するコードを⽣成する
ということを再帰的に⾏います。被演算⼦の命令列は、(construct-arglistに
よって)arglに引数リストを構築するコードと組み合わされ、結果となる引数
リストのコードは⼿続きコードと (compile-procedure-callによって⽣成さ
れる) ⼿続き呼び出しを実⾏するコードと組み合わされます。コード列を連結
するにあたって、envレジスタは演算⼦の評価の前後で保存する必要がありま
す (演算⼦の評価が envを変更するかもしれず、また envは被演算⼦の評価で
必要となるため)。また、procレジスタは引数リスト構築の前後で保存する必
要があります (被演算⼦を評価する際に、実際の⼿続き適⽤に必要な procレジ
スタを変更するかもしれないため)。continueも、⼿続き呼び出しのリンクに
必要となるため、全体を通して保存しなければなりません。
(define (compile-application exp target linkage)

(let ((proc-code (compile (operator exp) 'proc 'next))
(operand-codes
(map (lambda

(operand) (compile operand 'val 'next))
(operands exp))))

(preserving '(env continue)
proc-code
(preserving '(proc continue)
(construct-arglist operand-codes)
(compile-procedure-call target linkage)))))

引数リストを構築するコードは、それぞれの被演算⼦を評価して valに⼊れ、
それからその値を arglに集積中の引数リストに consします。arglに順に引
数を cons していくので、結果となるリストの中で引数が最初から最後という
順番になるように、最後の引数から始めて最初のもので終わるようにする必要
があります。この⼀連の評価のために arglを空リストに初期化して⼀命令を
無駄にする代わりに、最初のコード列に arglの初期値を構築させるようにし
ます。そのため、引数リスト構築の⼀般的な形式は以下のようになります。
⟨valをターゲットとして最後の被演算⼦をコンパイル⟩
(assign argl (op list) (reg val))
⟨valをターゲットとして次の被演算⼦をコンパイル⟩
(assign argl (op cons) (reg val) (reg argl))
. . .

⟨valをターゲットとして最初の被演算⼦をコンパイル⟩
(assign argl (op cons) (reg val) (reg argl))

arglは、(そこまでで集積した引数が失われないように)最初ひとつ以外の各
被演算⼦の評価の前後で保存する必要があります。また、envは (残りの被演算
⼦評価で使うために)最後のひとつ以外の各被演算⼦の評価の前後で保存する
必要があります。
この引数コードのコンパイルには少し注意が必要です。評価対象の最初の

被演算⼦を特別扱いすることになり、また arglと envを別々の位置で保存す
る必要があるからです。construct-arglist ⼿続きは個々の被演算⼦を評価す
るコードを引数として取ります。被演算⼦がなければ、単純に以下の命令を作
ります。
(assign argl (const ()))

被演算⼦がある場合は、construct-arglistは最後の引数で arglを初期化す
るコードを作り、引数の残りを評価してそれらを順次 arglにつなげるコード
を連結していきます。引数を最後から最初という順番で処理するために、被演
算⼦のコード列のリストを compile-applicationによって渡された順番の逆
順にする必要があります。

(define (construct-arglist operand-codes)
(let ((operand-codes (reverse operand-codes)))

(if (null? operand-codes)
(make-instruction-sequence '() '(argl)
'((assign argl (const ()))))

(let ((code-to-get-last-arg
(append-instruction-sequences
(car operand-codes)
(make-instruction-sequence '(val) '(argl)
'((assign argl (op list) (reg val)))))))

(if (null? (cdr operand-codes))
code-to-get-last-arg
(preserving '(env)
code-to-get-last-arg
(code-to-get-rest-args
(cdr operand-codes))))))))

(define (code-to-get-rest-args operand-codes)
(let ((code-for-next-arg

(preserving '(argl)
(car operand-codes)
(make-instruction-sequence '(val argl) '(argl)
'((assign argl

(op cons) (reg val) (reg argl)))))))
(if (null? (cdr operand-codes))

code-for-next-arg
(preserving '(env)
code-for-next-arg
(code-to-get-rest-args (cdr operand-codes))))))

⼿続きの適⽤
組み合わせの要素を評価した後に、コンパイルされたコードは proc内の⼿続
きを argl内の引数に適⽤しなければなりません。このコードは、本質的に4.1.1
節のメタ循環評価器の apply ⼿続きや、5.4.1 節の明⽰制御評価器の apply-
dispatch エントリポイントと同じディスパッチを実⾏します。まず、適⽤

する⼿続きが基本⼿続きか複合⼿続きか確認します。基本⼿続きには apply-
primitive-procedureを使います。これがどのようにコンパイルされた⼿続き
を扱うかについては、すぐ後で⾒ていくことにします。⼿続き適⽤のコードは
以下の形式を持ちます。
(test (op primitive-procedure?) (reg proc))
(branch (label primitive-branch))

compiled-branch
⟨与えられたターゲットと適切なリンクでコンパイル済み⼿続きを適⽤するコ
ード⟩
primitive-branch
(assign ⟨target⟩

(op apply-primitive-procedure)
(reg proc)
(reg argl))

⟨リンク⟩
after-call

コンパイル済み⼿続きの分岐 (compiled-branch)は基本⼿続きの分岐 (primitive-
branch)をスキップする必要があります。そのため、元の⼿続き呼び出しのリ
ンクが nextであった場合、複合⼿続きの分岐は基本⼿続きの分岐の後に挿⼊
されたラベルにジャンプするリンクを使う必要があります (これは、compile-
ifの中で真の分岐に使われるリンクと似ています)。
(define (compile-procedure-call target linkage)

(let ((primitive-branch (make-label 'primitive-branch))
(compiled-branch (make-label 'compiled-branch))
(after-call (make-label 'after-call)))

(let ((compiled-linkage
(if (eq? linkage 'next) after-call linkage)))

(append-instruction-sequences
(make-instruction-sequence '(proc) '()
`((test (op primitive-procedure?) (reg proc))

(branch (label ,primitive-branch))))
(parallel-instruction-sequences
(append-instruction-sequences
compiled-branch

(compile-proc-appl target compiled-linkage))
(append-instruction-sequences
primitive-branch
(end-with-linkage linkage
(make-instruction-sequence '(proc argl)

(list target)
`((assign ,target

(op apply-primitive-procedure)
(reg proc)
(reg argl)))))))

after-call))))

基本⼿続きと複合⼿続きの分岐は、compile-ifでの真と偽の分岐と同じよ
うに、通常の append-instruction-sequencesではなく parallel-instruction-
sequencesを使って追加されます。これらは順番に実⾏されるわけではないか
らです。

コンパイル済み⼿続きの適⽤
⼿続きの適⽤を扱うコードは、⽣成する命令列はとても短いのですが、コンパ
イラの中で最も難しい部分です。(compile-lambdaに構築された)コンパイル
済み⼿続きは、⼿続きのコードが始まる場所を指定するラベルであるエントリ
ポイントを持っています。このエントリポイントのコードは結果を計算して
valに⼊れ、(goto (reg continue))という命令を実⾏して戻ります。そのた
め、与えられたターゲットとリンクによるコンパイル済み⼿続きの適⽤コード
(これは compile-proc-applによって⽣成されます) は、リンクがラベルであ
れば以下のようになり、
(assign continue (label proc-return))
(assign val (op compiled-procedure-entry) (reg proc))
(goto (reg val))

proc-return
(assign ⟨target⟩ (reg val)) ; ターゲットが val でない場合に挿⼊
される
(goto (label ⟨リンク⟩)) ; リンクコード
リンクが returnであれば以下のようになると期待できます。

(save continue)
(assign continue (label proc-return))
(assign val (op compiled-procedure-entry) (reg proc))
(goto (reg val))

proc-return
(assign ⟨target⟩ (reg val)) ; ターゲットが val でない場合に挿⼊
される
(restore continue)
(goto (reg continue)) ; リンクコード
このコードは、⼿続きが proc-returnに戻るように continueを設定し、⼿続
きのエントリポイントにジャンプします。proc-returnのコードは、(必要であ
れば)⼿続きの結果を valからターゲットレジスタに転送し、それからリンク
によって指定された位置にジャンプします (リンクは常に returnかラベルで
す。compile-procedure-callは、複合⼿続きの分岐の nextリンクを after-
callラベルに置き換えるからです)。
実際、ターゲットが valでない場合、これがまさにコンパイラが⽣成する

コードとなります。39 しかし、通常はターゲットは valであるため (コンパイ
ラが別のレジスタを指定するのは演算⼦の評価のターゲットを procにすると
きだけです)、⼿続きの結果は直接ターゲットレジスタに⼊れられることにな
るので、それをコピーする特別な場所に戻る必要はありません。そのため、そ
の場所に戻る代わりに、呼び出し元のリンクで指定された場所に⼿続きが直接
“戻る”ように continueを設定して、コードを単純化します。
⟨continueにリンクを設定⟩
(assign val (op compiled-procedure-entry) (reg proc))
(goto (reg val))

リンクがラベルであれば、⼿続きがそのラベルに戻るように continueを設定
します (つまり、⼿続きの終端の (goto (reg continue)) が、上記の proc-
returnの (goto (label <linkage>))と等価になります)。
(assign continue (label ⟨リンク⟩))
(assign val (op compiled-procedure-entry) (reg proc))
(goto (reg val))

39実際のところ、ターゲットが valでなく、リンクが returnである場合にはエラーを
発⽣させます。returnリンクを要求する場所は⼿続きのコンパイルの中だけで、⼿続き
は値を valに⼊れる約束だからです。

もしリンクが returnなら、continueを設定する必要はありません。continue
にはすでに望ましい位置が⼊っています (これは、⼿続きの終端の (goto (reg
continue))が、proc-returnの (goto (reg continue))が⾏くはずの場所に
直接⾏くということです)。
(assign val (op compiled-procedure-entry) (reg proc))
(goto (reg val))

この returnリンクの実装で、コンパイラは末尾再帰のコードを⽣成します。⼿
続きの本体の最後のステップとして⼿続き呼び出しを⾏うと、スタックに何の
情報も保存せず、直接移動を⾏うことになります。
このようにする代わりに、リンクが returnでターゲットが valである場合

を上に⽰した val以外のターゲットと同じように扱うと仮定してみましょう。
こうすると、末尾再帰を壊してしまいます。このシステムは、どんな式に対し
ても相変わらず同じ値を返すでしょう。しかし、⼿続きを呼ぶたびに continue
を保存することになり、呼び出しから帰ってきて (必要のない)保存の取り消
しをすることになります。この余分な保存は、⼊れ⼦の⼿続き呼び出しの間に
蓄積されていきます。40

compile-proc-applは、呼び出しのターゲットが valかどうか、リンクが
returnかどうかによって、四つの場合を考慮して上記の⼿続き適⽤コードを
⽣成します。命令列は、すべてのレジスタを変更するように宣⾔されています。
これは、⼿続きの本体を実⾏すると、どのレジスタがどう変更されるかわから
ないからです。41 また、ターゲットが valで、リンクが returnの場合のコー

40コンパイラに末尾再帰のコードを⽣成させるというのは素直な考え⽅のように思え
るかもしれません。しかし、C⾔語や Pascalを含め、⼀般的な⾔語ではこれを⾏わず、
そのためこれらの⾔語では反復プロセスを⼿続き呼び出しだけで表現することはできま
せん。これらの⾔語で末尾再帰が難しいのは、それらの実装ではスタックをリターンア
ドレスを格納するのに使うだけでなく、⼿続きの引数や局所変数を格納するためにも使
っているからです。この本で記述されている Schemeの実装は、引数と変数をメモリに
⼊れ、ガベージコレクションの対象にしています。変数と引数にスタックを使う理由は、
ほかのところでガベージコレクションを使わない⾔語では、そうすることによってガベ
ージコレクションが必要なくなり、またそれがより効率的だと⼀般に信じられていると
いうことです。実際のところ、⾼機能な Lispコンパイラは、末尾再帰を壊さずに引数の
ためにスタックを使うことができます (この記述についてはHanson 1990を参照してく
ださい)。また、そもそもスタック割り当てはガベージコレクションより効率的なのかと
いうところにも議論がありますが、この問題の詳細はコンピュータアーキテクチャの細
部によるようです (この問題に反対の⽴場からの意⾒については、Appel 1987とMiller
and Rozas 1994を参照してください)。

41変数 all-regsは、全レジスタの名前のリストに束縛されています。

ド列は continue を必要とすると宣⾔されていることに気をつけてください。
continueは⼆命令の列では明⽰的に使われていませんが、コンパイル済み⼿
続きに⼊るときには continueが正しい値を持っているということを保証する
必要があります。
(define (compile-proc-appl target linkage)

(cond ((and (eq? target 'val) (not (eq? linkage 'return)))
(make-instruction-sequence '(proc) all-regs

`((assign continue (label ,linkage))
(assign val (op compiled-procedure-entry)

(reg proc))
(goto (reg val)))))

((and (not (eq? target 'val))
(not (eq? linkage 'return)))

(let ((proc-return (make-label 'proc-return)))
(make-instruction-sequence '(proc) all-regs
`((assign continue (label ,proc-return))

(assign val (op compiled-procedure-entry)
(reg proc))

(goto (reg val))
,proc-return
(assign ,target (reg val))
(goto (label ,linkage))))))

((and (eq? target 'val) (eq? linkage 'return))
(make-instruction-sequence
'(proc continue)
all-regs
'((assign val (op compiled-procedure-entry)

(reg proc))
(goto (reg val)))))

((and (not (eq? target 'val))
(eq? linkage 'return))

(error "return linkage, target not val: COMPILE"
target))))

(define all-regs '(env proc val argl continue))

5.5.4 命令列のコンパイル
この節では、命令列の表現⽅法と組み合わせ⽅の詳細を述べます。命令列は必
要とされるレジスタ、変更されるレジスタ、実際の命令というリストとして表
現されるという5.5.1節の記述を思い出してください。このほかに、ラベル (記
号)についても、レジスタを必要としたり変更したりしない単純化した命令列
と考えることにします。よって、命令列によって必要とされたり修正されたり
するレジスタを決定するには、以下のセレクタを使います。
(define (registers-needed s)

(if (symbol? s) '() (car s)))
(define (registers-modified s)

(if (symbol? s) '() (cadr s)))
(define (statements s)

(if (symbol? s) (list s) (caddr s)))

また、与えられた命令列が与えられたレジスタを必要としたり変更したりする
かを判断するには、以下の述語を使います。
(define (needs-register? seq reg)

(memq reg (registers-needed seq)))
(define (modifies-register? seq reg)

(memq reg (registers-modified seq)))

これらの述語とセレクタを使って、コンパイラを全体を通して使⽤される、命
令列に対するさまざまな結合器を実装できます。
基本的な結合器は append-instruction-sequencesです。これは、順に実

⾏される任意の数の命令列を引数として取り、すべての命令列の命令⽂を連結
した命令⽂を持つ命令列を返します。難しいところは、結果となる命令列が必
要とするレジスタと変更するレジスタを決定するというところです。命令列の
いずれかが変更するレジスタは、変更するレジスタとなります。必要とするレ
ジスタは、最初の命令列が実⾏される前に初期化されなければならないレジス
タ (最初の命令列に必要とされるレジスタ)と、それに加えて、そのほかの命令
列に必要とされるレジスタのうちそれ以前の命令列によって初期化 (変更)さ
れていないものです。
命令列は append-2-sequencesによって⼀度に⼆つが接続されます。これ

は⼆つの命令列 seq1と seq2を取り、seq1の命令⽂の後に seq2の命令⽂が置
かれた命令⽂を持つ命令列を返します。結果となる命令列の変更するレジスタ
は、seq1か seq2のいずれかが変更するレジスタになります。また、この命令

列の必要とするレジスタは、seq1によって必要とされるレジスタに、seq2で
必要とされ、かつ seq1で変更されないレジスタを加えたものです (集合演算で
考えると、必要なレジスタの新しい集合は、seq1に必要とされるレジスタの集
合と、seq2に必要とされるレジスタと seq1に変更されるレジスタの差集合と
の、和集合ということになります)。
(define (append-instruction-sequences . seqs)

(define (append-2-sequences seq1 seq2)
(make-instruction-sequence
(list-union
(registers-needed seq1)
(list-difference (registers-needed seq2)

(registers-modified seq1)))
(list-union (registers-modified seq1)

(registers-modified seq2))
(append (statements seq1) (statements seq2))))

(define (append-seq-list seqs)
(if (null? seqs)

(empty-instruction-sequence)
(append-2-sequences
(car seqs)
(append-seq-list (cdr seqs)))))

(append-seq-list seqs))

この⼿続きは、リストとして表現された集合を操作する簡単な演算をいくつか
使います。この表現は、2.3.3節で述べた (順序なし)集合表現に似ています。
(define (list-union s1 s2)

(cond ((null? s1) s2)
((memq (car s1) s2) (list-union (cdr s1) s2))
(else (cons (car s1) (list-union (cdr s1) s2)))))

(define (list-difference s1 s2)
(cond ((null? s1) '())

((memq (car s1) s2) (list-difference (cdr s1) s2))
(else (cons (car s1)

(list-difference (cdr s1) s2)))))

⼆番⽬に重要な命令列結合器である preservingは、レジスタリスト regsと、
順に実⾏する⼆つの命令列 seq1と seq2を取ります。この結合器の返す命令

列の持つ命令⽂は、seq1 の命令⽂に seq2 の命令⽂が続くものですが、regs
のうち seq1に変更されるが seq2に必要とされるレジスタを守るため、seq1
の前後には適切な save と restore が追加されます。これを達成するために、
preservingはまず、必要な save、seq1の命令⽂、必要な restoreを持つ命令
列を作ります。この命令列は、seq1によって必要とされるレジスタに加えて、
保存と復元が⾏われるレジスタを必要とし、seq1で変更されるレジスタのう
ち、保存と復元が⾏われるレジスタを除いたものを変更します。次に、この拡
張した命令列と seq2が通常の⽅法で連結されます。次の⼿続きはこの戦略を
再帰的に実装したもので、保存するべきレジスタのリストを下っていきます。
42

(define (preserving regs seq1 seq2)
(if (null? regs)

(append-instruction-sequences seq1 seq2)
(let ((first-reg (car regs)))

(if (and (needs-register? seq2 first-reg)
(modifies-register? seq1 first-reg))

(preserving (cdr regs)
(make-instruction-sequence
(list-union (list first-reg)

(registers-needed seq1))
(list-difference (registers-modified seq1)

(list first-reg))
(append `((save ,first-reg))

(statements seq1)
`((restore ,first-reg))))

seq2)
(preserving (cdr regs) seq1 seq2)))))

命令列結合器 tack-on-instruction-sequenceは、compile-lambdaが⼿続き
の本体をほかの命令列に連結するために使うものです。⼿続きの本体が “埋め
込まれている”のは、組み合わせた命令列の⼀部として実⾏するためではない
ので、それのレジスタ使⽤は埋め込む命令列のレジスタ使⽤に影響しません。
そのため、⼿続きの本体をほかの命令列にくっつけるときには、それが必要と

42preservingが appendを三つの引数で呼び出していることに注意してください。こ
の本で⽰した appendの定義は⼆つの引数しか受け⼊れませんが、Schemeは任意の数の
引数を取る append⼿続きを標準的に提供しています。

するレジスタと変更するレジスタの集合を無視します。
(define (tack-on-instruction-sequence seq body-seq)

(make-instruction-sequence
(registers-needed seq)
(registers-modified seq)
(append (statements seq)

(statements body-seq))))

compile-ifと compile-procedure-callは、テストに続く⼆つの選択肢の分
岐を追加するために、parallel-instruction-sequencesという特別な結合器
を使います。⼆つの分岐が順に実⾏されることは決してありません。テストの
評価がどのようなものであっても、どちらかひとつの分岐に⼊ることになりま
す。このため、⼆つ⽬の分岐に必要とされるレジスタは、それが⼀つ⽬の分岐
に修正されるものであっても、組み合わせ後の命令列で必要とされることにな
ります。
(define (parallel-instruction-sequences seq1 seq2)

(make-instruction-sequence
(list-union (registers-needed seq1)

(registers-needed seq2))
(list-union (registers-modified seq1)

(registers-modified seq2))
(append (statements seq1)

(statements seq2))))

5.5.5 コンパイル済みコードの例
ここまでで、コンパイラのすべての要素について⾒てきたことになります。こ
こでは、コンパイル済みコードの例を詳しく調べて、要素がどのように組み合
わされているのか⾒てみましょう。ここでは、compileを呼ぶことによって、
再帰 factorial⼿続きの定義をコンパイルすることにします。
(compile
'(define (factorial n)

(if (= n 1)
1
(* (factorial (- n 1)) n)))

646



'val
'next)

define式の値はレジスタ valに⼊れるよう規定しています。defineを実⾏し
た後にコンパイル済みコードが何をするかには興味がないので、リンクとして
nextを選んでいることに特に意味はありません。

compileは、式が定義であると判断し、compile-definitionを呼び出して、
前から順に (ターゲット valに)割り当てる値を求めるコード、定義を組み込
むコード、defineの値 (記号 ok)をターゲットレジスタに⼊れるコード、リン
クコードという並びをコンパイルします。envは定義を組み込むのに必要なの
で、値の計算の前後で保存されます。リンクは nextなので、この場合リンク
コードはありません。よって、コンパイル済みコードの⾻組みは次のようにな
ります。
⟨値を計算するコードで変更されるなら envを保存⟩
⟨ターゲット val、リンク nextでの定義の値のコンパイル⟩
⟨上で保存していたら envを復元⟩
(perform (op define-variable!)

(const factorial)
(reg val)
(reg env))

(assign val (const ok))

変数 factorialの値を作るためにコンパイルする式は lambda式で、その値は
階乗を計算する⼿続きです。compileは compile-lambdaを呼んでこれを処理
します。compile-lambdaは⼿続きの本体をコンパイルし、それに新しいエン
トリポイントとしてラベルをつけ、新しいエントリポイントにある⼿続き本体
を実⾏時環境と組み合わせて結果を valに割り当てる命令を⽣成します。次に、
命令列はこの時点で挿⼊されるコンパイル済みコードをスキップします。⼿続
きのコード⾃⾝は、まず⼿続きの定義環境を、仮引数 nを⼿続きの引数に束縛
するフレームによって拡張します。それから実際の⼿続き本体が続きます。変
数の値のためのこのコードは envレジスタを変更しないので、上で⽰したオプ
ションの saveと restoreは⽣成されません (entry2にある⼿続きのコードは
この時点では実⾏されません。そのため、それが envを使うかどうかは影響し
ません)。よって、コンパイル済みコードの⾻組みは次のようになります。

(assign val
(op make-compiled-procedure)

(label entry2)
(reg env))

(goto (label after-lambda1))
entry2

(assign env (op compiled-procedure-env) (reg proc))
(assign env

(op extend-environment)
(const (n))
(reg argl)
(reg env))

⟨⼿続き本体のコンパイル⟩
after-lambda1

(perform (op define-variable!)
(const factorial)
(reg val)
(reg env))

(assign val (const ok))

⼿続きの本体は、常にターゲット val とリンク return を持つ命令列として
(compile-lambda-bodyによって)コンパイルされます。今回の場合、命令列は
ひとつの if式からなります。
(if (= n 1)

1
(* (factorial (- n 1)) n))

compile-ifが⽣成するコードは、まず最初に述語を (ターゲットを valとし
て)計算し、それからその結果を確認して、述語が偽であれば真の分岐をスキ
ップします。envと continueは述語コードの前後で保存されます。これらは
if式の残りの部分で必要となる可能性があるためです。この if式は⼿続きの
本体を構成する命令列内の最後の (そして唯⼀の)式であり、ターゲットは val
で、リンクは returnであるため、真と偽の分岐はどちらもターゲット valと
リンク returnでコンパイルされます (つまり、条件⽂の値が⼿続きの値となる
ということです。条件⽂の値はいずれかの分岐で計算される値です)。
⟨述語により変更され、分岐により必要とされるのであれば continue, envを
保存する⟩
⟨ターゲット val, リンク nextでの述語のコンパイル⟩

⟨上で保存していれば continue, envを復元する⟩
(test (op false?) (reg val))
(branch (label false-branch4))

true-branch5
⟨ターゲット val, リンク returnでの真の分岐のコンパイル⟩

false-branch4
⟨ターゲット val, リンク returnでの偽の分岐のコンパイル⟩

after-if3

述語 (= n 1)は⼿続き呼び出しです。これは演算⼦ (記号 =)を検索し、その値
を procに⼊れます。次に、引数である 1と、変数 nの値を arglに集めます。
それから、procに⼊っているものが基本⼿続きか複合⼿続きかをテストし、そ
れに従って基本⼿続きの分岐か複合⼿続きの分岐にディスパッチします。どち
らの分岐もラベル after-callから再開します。演算⼦と被演算⼦の評価の前
後ではレジスタを保存する必要がありますが、今回の場合はこれらの評価は問
題となるレジスタを変更しないため、結果的にレジスタの保存は⾏われません。

(assign proc
(op lookup-variable-value) (const =) (reg env))

(assign val (const 1))
(assign argl (op list) (reg val))
(assign val

(op lookup-variable-value) (const n) (reg env))
(assign argl (op cons) (reg val) (reg argl))
(test (op primitive-procedure?) (reg proc))
(branch (label primitive-branch17))

compiled-branch16
(assign continue (label after-call15))
(assign val (op compiled-procedure-entry) (reg proc))
(goto (reg val))

primitive-branch17
(assign val

(op apply-primitive-procedure)
(reg proc)
(reg argl))

after-call15

真の分岐は定数 1であり、これを (ターゲット valとリンク returnで)コンパ
イルすると、以下のようになります。
(assign val (const 1))
(goto (reg continue))

偽の分岐のコードはまた別の⼿続き呼び出しです。⼿続きは記号 *の値で、引
数は nと、また別の⼿続き呼び出し (factorialの呼び出し)の結果です。こ
れらの呼び出しは、それぞれ procと arglと、その⼿続き⾃⾝の基本⼿続き
と複合⼿続きの分岐を準備します。図 5.17は、⼿続き factorialの定義の完全
なコンパイル結果を⽰したものです。述語の前後で、⽣成される可能性のある
continueと envの saveと restoreが実際に⽣成されていることに注意して
ください。これらのレジスタは述語内の⼿続き呼び出しで変更され、分岐内の
⼿続き呼び出しと returnのリンクに必要とされるためです。

練習問題 5.33: 次の階乗⼿続きの定義について考えよ。これは上
に挙げたものとわずかに異なっている。
(define (factorial-alt n)

(if (= n 1)
1
(* n (factorial-alt (- n 1)))))

この⼿続きをコンパイルし、その結果のコードを factorialに対
して⽣成されるコードと⽐較せよ。⾒つけた違いについて説明せ
よ。⼆つのプログラムの実⾏効率に違いはあるだろうか。
練習問題 5.34: 反復階乗⼿続きをコンパイルせよ。
(define (factorial n)

(define (iter product counter)
(if (> counter n)

product
(iter (* counter product)

(+ counter 1))))
(iter 1 1))

結果のコードに注釈をつけ、factorialの反復版と再帰版のコー
ドで、⽚⽅がスタック空間を積み上げていきもう⽚⽅が⼀定のス
タック空間で動作するという違いを⽣む、本質的な違いを⽰せ。

図 5.17: ↓ factorial⼿続き定義のコンパイル結果
;; ⼿続きを構築し、⼿続き本体のコードをスキップする

(assign val
(op make-compiled-procedure)
(label entry2)
(reg env))

(goto (label after-lambda1))
entry2 ; factorialへの呼び出しはここから⼊ることになる

(assign env (op compiled-procedure-env) (reg proc))
(assign env

(op extend-environment)
(const (n))
(reg argl)
(reg env))

;; 実際の⼿続き本体開始
(save continue)
(save env)

;; (= n 1)を計算
(assign proc

(op lookup-variable-value)
(const =)
(reg env))

(assign val (const 1))
(assign argl (op list) (reg val))
(assign val

(op lookup-variable-value)
(const n)
(reg env))

(assign argl (op cons) (reg val) (reg argl))
(test (op primitive-procedure?) (reg proc))
(branch (label primitive-branch17))

compiled-branch16
(assign continue (label after-call15))
(assign val (op compiled-procedure-entry) (reg proc))
(goto (reg val))

primitive-branch17
(assign val

(op apply-primitive-procedure)
(reg proc)
(reg argl))

after-call15 ; valには (= n 1)の結果が⼊っている

(restore env)
(restore continue)
(test (op false?) (reg val))
(branch (label false-branch4))

true-branch5 ; return 1
(assign val (const 1))
(goto (reg continue))

false-branch4
;; (* (factorial (- n 1)) n)を計算して返す

(assign proc
(op lookup-variable-value)
(const *)
(reg env))

(save continue)
(save proc) ; *⼿続きを保存する
(assign val

(op lookup-variable-value)
(const n)
(reg env))

(assign argl (op list) (reg val))
(save argl) ; *の部分引数リストを保存

;; *のもうひとつの引数である (factorial (- n 1))を計算
(assign proc

(op lookup-variable-value)
(const factorial)
(reg env))

(save proc) ; factorial⼿続きを保存
;; factorialの引数である (- n 1)を計算

(assign proc
(op lookup-variable-value)
(const -)
(reg env))

(assign val (const 1))
(assign argl (op list) (reg val))
(assign val

(op lookup-variable-value)
(const n)
(reg env))

(assign argl (op cons) (reg val) (reg argl))
(test (op primitive-procedure?) (reg proc))
(branch (label primitive-branch8))

compiled-branch7

(assign continue (label after-call6))
(assign val (op compiled-procedure-entry) (reg proc))
(goto (reg val))

primitive-branch8
(assign val

(op apply-primitive-procedure)
(reg proc)
(reg argl))

after-call6 ; valには (- n 1)の結果が⼊っている
(assign argl (op list) (reg val))
(restore proc) ; factorialを復元

;; factorialを適⽤
(test (op primitive-procedure?) (reg proc))
(branch (label primitive-branch11))

compiled-branch10
(assign continue (label after-call9))
(assign val (op compiled-procedure-entry) (reg proc))
(goto (reg val))

primitive-branch11
(assign val

(op apply-primitive-procedure)
(reg proc)
(reg argl))

after-call9 ; valには (factorial (- n 1))の結果が⼊ってい
る

(restore argl) ; *の部分引数リストを復元
(assign argl (op cons) (reg val) (reg argl))
(restore proc) ; *を復元
(restore continue)

;; *を適⽤し、その値を返す
(test (op primitive-procedure?) (reg proc))
(branch (label primitive-branch14))

compiled-branch13
;; ここの複合⼿続きは末尾再帰で呼ばれることに注意すること

(assign val (op compiled-procedure-entry) (reg proc))
(goto (reg val))

primitive-branch14
(assign val

(op apply-primitive-procedure)
(reg proc)
(reg argl))

(goto (reg continue))

after-call12
after-if3
after-lambda1
;; ⼿続きを変数 factorialに割り当てる

(perform (op define-variable!)
(const factorial)
(reg val)
(reg env))

(assign val (const ok))

練習問題 5.35: 図 5.18に⽰すコードは、どのような式をコンパイ
ルして⽣成されたものか。
図 5.18: ↓ コンパイラ出⼒の例。練習問題 5.35参照。
(assign val

(op make-compiled-procedure)
(label entry16)
(reg env))

(goto (label after-lambda15))
entry16

(assign env (op compiled-procedure-env) (reg proc))
(assign env

(op extend-environment)
(const (x))
(reg argl)
(reg env))

(assign proc
(op lookup-variable-value)
(const +)
(reg env))

(save continue)
(save proc)
(save env)
(assign proc

(op lookup-variable-value)
(const g)
(reg env))

(save proc)
(assign proc

(op lookup-variable-value)
(const +)

(reg env))
(assign val (const 2))
(assign argl (op list) (reg val))
(assign val

(op lookup-variable-value)
(const x)
(reg env))

(assign argl (op cons) (reg val) (reg argl))
(test (op primitive-procedure?) (reg proc))
(branch (label primitive-branch19))

compiled-branch18
(assign continue (label after-call17))
(assign val (op compiled-procedure-entry) (reg proc))
(goto (reg val))

primitive-branch19
(assign val

(op apply-primitive-procedure)
(reg proc)
(reg argl))

after-call17
(assign argl (op list) (reg val))
(restore proc)
(test (op primitive-procedure?) (reg proc))
(branch (label primitive-branch22))

compiled-branch21
(assign continue (label after-call20))
(assign val (op compiled-procedure-entry) (reg proc))
(goto (reg val))

primitive-branch22
(assign val

(op apply-primitive-procedure)
(reg proc)
(reg argl))

after-call20
(assign argl (op list) (reg val))
(restore env)
(assign val

(op lookup-variable-value)
(const x)
(reg env))

(assign argl (op cons) (reg val) (reg argl))
(restore proc)

(restore continue)
(test (op primitive-procedure?) (reg proc))
(branch (label primitive-branch25))

compiled-branch24
(assign val

(op compiled-procedure-entry)
(reg proc))

(goto (reg val))
primitive-branch25

(assign val
(op apply-primitive-procedure)
(reg proc)
(reg argl))

(goto (reg continue))
after-call23
after-lambda15

(perform (op define-variable!)
(const f)
(reg val)
(reg env))

(assign val (const ok))

練習問題 5.36: 我々のコンパイラは、組み合わせの被演算⼦に対
してどのような評価順を⽣成するだろうか。左から右だろうか、右
から左だろうか、またはほかの順序だろうか。この順序はコンパ
イラのどこで決定されるだろうか。コンパイラを修正し、別の評
価順を⽣成するようにせよ (5.4.1節での明⽰制御評価器の評価順
の議論参照)。被演算⼦の評価順を変えると、引数リストを構築す
るコードの効率にどのような影響があるだろうか。
練習問題 5.37: スタック使⽤を最適化するためのコンパイラの
preserving のメカニズムを理解する⽅法のひとつとして、この
考えを使わなかったとしたらどのような演算が余分に⽣成されて
いたかを⾒るというものがある。preserving を修正し、save と
restore演算を常に⽣成するようにせよ。簡単な式をいくつかコ
ンパイルし、⽣成される不必要なスタック演算を特定せよ。⽣成
されたコードと、preservingメカニズムをそのままにした状態で
⽣成されるコードとを⽐較せよ。
練習問題 5.38: 我々のコンパイラは、不必要なスタック演算を避
けることに関してはうまくやっているが、⾔語の基本⼿続きの呼

び出しをマシンによって提供される基本演算を使ってコンパイル
することに関しては、まったくうまくやっていない。例えば、(+ a
1)を求めるためにどれだけのコードがコンパイルされるか考えて
みよう。このコードは、引数リストを arglに準備し、(環境内で記
号 +を探すことによって⾒つける)基本加算⼿続きを procに⼊れ、
この⼿続きが基本⼿続きか複合⼿続きかテストする。コンパイラ
は、このテストを⾏うコードに加えて、基本⼿続きと複合⼿続き
の分岐 (そのうちひとつしか実⾏されない)も常に⽣成する。コン
トローラの中で基本⼿続きを実装する部分は⽰さなかったが、そ
れらの命令はマシンのデータパスの基本算術演算を使うものと仮
定する。ここで、もしコンパイラが基本⼿続きを オープンコード
(open-code)できたら—つまり、これらの基本マシン演算を直接使
うコードを⽣成できたら—⽣成されるコードがどれだけ少なくな
るか考えてみよう。式 (+ a 1)は、以下のような単純なものにコ
ンパイルされることになるかもしれない。43

(assign
val (op lookup-variable-value) (const a) (reg env))

(assign val (op +) (reg val) (const 1))

この課題では、我々のコンパイラを拡張し、選ばれた基本⼿続き
のオープンコードをサポートするようにする。それらの基本⼿続
きの呼び出しに対しては、⼀般的な⼿続き適⽤のコードではなく、
特別⽬的のコードが⽣成される。これをサポートするために、マ
シンには arg1と arg2という特別な引数レジスタを追加する。マ
シンの基本数値演算は、⼊⼒を arg1と arg2から取る。その結果
は、val, arg1, arg2のいずれかに⼊れる。
コンパイラは、ソースプログラム内のオープンコードな基本⼿続
きの適⽤を認識できる必要がある。compile⼿続きにディスパッ
チを追加し、いま認識できる予約語 (特殊形式)に加えてそれらの
名前も認識できるようにする。44 特殊形式それぞれに対して、私

43ここでは、+という同じ記号をソース⾔語の⼿続きとマシン演算の両⽅を⽰すために
使っています。⼀般的には、ソース⾔語の基本⼿続きとマシンの基本演算の間に⼀対⼀
の対応はありません。

44基本⼿続きを予約語に⼊れるというのは⼀般的にはよくない考えです。そうしてし
まうと、ユーザーがそれらの名前を別の⼿続きに束縛し直すことができなくなるからで
す。そのうえ、いま使われているコンパイラに予約語を追加すると、それらの名前で⼿

たちのコンパイラはコード⽣成器を持っている。この練習問題で
は、オープンコードの基本⼿続きに対して、コード⽣成器を構築
する。

a. オープンコードの基本⼿続きはすべて、特殊形式とは違って、
被演算⼦の評価を必要とする。オープンコードのコード⽣成
器すべてによって使われるコード⽣成器 spread-arguments
を書け。spread-argumentsは被演算⼦のリストを取り、連
続した引数レジスタをターゲットとして与えられた被演算⼦
をコンパイルする。被演算⼦はオープンコードの基本⼿続き
の呼び出しを含む可能性があるため、被演算⼦評価の間に引
数レジスタを保存しなければならないことに注意せよ。

b. 基本⼿続き =, *, -, +のそれぞれについてコード⽣成器を書
け。コード⽣成器は、演算⼦を含む組み合わせ、ターゲット、
リンクを引数に取り、引数をレジスタに⼊れ与えられたター
ゲットと与えられたリンクで演算を実⾏するコードを⽣成す
る。⼆つの被演算⼦を持つ式が扱えればそれでよい。compile
がこれらのコード⽣成器にディスパッチを⾏うようにせよ。

c. 新しいコンパイラを factorialの例で試せ。結果のコードを
オープンコードなしで⽣成した結果と⽐較せよ。

d. +と *のコード⽣成器を拡張し、任意の数の被演算⼦を持つ
式を扱えるようにせよ。三つ以上の被演算⼦を持つ式は、そ
れぞれが⼆つだけの⼊⼒を持つ演算の列にコンパイルする必
要があるだろう。

5.5.6 レキシカルアドレッシング
コンパイラによって⼀般的に⾏われる最適化のひとつとして、変数検索の
最適化があります。ここまで実装してきたコンパイラは、評価器の lookup-
variable-value演算を使うコードを⽣成します。この演算は、実⾏時環境を
外に向かってフレームごとに⾒ていきながら、現在束縛されているそれぞれの
変数と⽐較することによって変数を検索します。この検索は、フレームの⼊れ
⼦が深かったり、変数が多かったりすると、⾼コストになりえます。例えば、
続きを定義している既存のプログラムが動作しなくなります。この問題を避けるための
考え⽅については練習問題 5.44を参照してください。

以下の式によって返される⼿続きの適⽤で、式 (* x y z)を評価する途中に x
の値を検索するという問題について考えみましょう。
(let ((x 3) (y 4))

(lambda (a b c d e)
(let ((y (* a b x)) (z (+ c d x)))

(* x y z))))

let式は lambda組み合わせのシンタックスシュガーに過ぎないため、この式
は以下と等価です。
((lambda (x y)

(lambda (a b c d e)
((lambda (y z) (* x y z))
(* a b x)
(+ c d x))))

3
4)

lookup-variable-valueが xを検索するたびに、記号 xが (⼀つ⽬のフレーム
では) y, zや (⼆つ⽬のフレームでは)a, b, c, d, eと eq?でないことを判定する
必要があります。ここではさしあたり、プログラムは defineを使わず、変数
は lambdaによってのみ束縛されるものと仮定します。私たちの⾔語はレキシ
カルスコープを持つ⾔語なので、式の実⾏時環境は、その式があるプログラム
のレキシカルな構造を反映した構造を持ちます。45 そのため、コンパイラが上
の式を分析するとき、⼿続きが適⽤されるときはいつでも、(* x y z)の変数
xが現在のフレームから⼆つ外のフレームで⾒つかるということ、またそのフ
レームの最初の変数であるということを知ることができます。

lexical-address-lookupという新しい種類の変数検索演算を作ることで、
この事実を活⽤することができます。この演算は、環境と、⼆つの数値からな
るレキシカルアドレス (lexical address)を引数として取ります。⼆つの数値は、
フレームをいくつパスするかを指定するフレーム数 (frame number)と、そのフ
レーム内で変数をいくつパスするかを指定する変位数 (displacement number)
です。lexical-address-lookupは、現在のフレームに対してその相対的なレ
キシカルアドレスの位置に格納されている変数の値を返します。私たちのマシ
ンに lexical-address-lookup演算を追加すると、lookup-variable-valueを

45内部定義を許すのであれば、それらを取り除かない限り、これは成り⽴ちません。練
習問題 5.43参照。

使う代わりにこの演算を使って変数を参照するコードをコンパイラに⽣成させ
ることができます。同様に、コンパイルされたコードは set-variable-value!
の代わりに新しい lexical-address-set!演算を使うことができます。
そのようなコードを⽣成するためには、変数の参照をコンパイルしようと

するとき、コンパイラはその変数のレキシカルアドレスを確定できなければな
りません。プログラム中の変数のレキシカルアドレスは、それがコードのどこ
にあるのかに依存します。例えば、以下のプログラムで、式 ⟨e1⟩の中では、x
アドレスは (2, 0)、つまり 2フレーム後ろで、そのフレームの最初の変数とい
うことになります。その位置では、yはアドレス (0, 0)であり、cはアドレス
(1, 2)です。式 ⟨e2 ⟩では、xは (1, 0)、yは (1, 1)、cは (0, 2)です。
((lambda (x y)

(lambda (a b c d e)
((lambda (y z) ⟨e1⟩)
⟨e2⟩
(+ c d x))))

3
4)

コンパイラがレキシカルアドレッシングを使うコードを⽣成するひとつの⽅
法として、コンパイル時環境 (compile-time environment)と呼ばれるデータ構
造を管理するというものがあります。このデータ構造は、特定のアクセス演算
が実⾏されるときにどの変数が実⾏時環境のどのフレームのどの位置にある
かということを記録するものです。コンパイル時環境はフレームのリストで、
それぞれが変数リストを持っています (もちろん、変数に束縛される値はあり
ません。コンパイル時には値は計算されないからです)。コンパイル時環境は
compile の追加の引数になり、それぞれのコード⽣成器に渡されます。トッ
プレベルでの compileへの呼び出しには、空のコンパイル時環境を使います。
lambdaの本体がコンパイルされるときは、compile-lambda-bodyは⼿続きの
仮引数を持つフレームによってコンパイル時環境を拡張し、本体を構成する命
令列がその拡張された環境を引数としてコンパイルされるようにします。コン
パイルの各時点で、compile-variableと compile-assignmentは適切なレキ
シカルアドレスを⽣成するためにコンパイル時環境を使⽤します。
練習問題 5.39から練習問題 5.43は、コンパイラにレキシカルな検索を組み

込むために、このレキシカルアドレッシング戦略の概要をどのように完成させ
るべきかを述べたものです。練習問題 5.44では、コンパイル時環境の別の使い
⽅を記述します。

練習問題 5.39: 新しい検索演算を実装する lexical-address-
lookup⼿続きを書け。引数はレキシカルアドレスと実⾏時環境の
⼆つで、指定したレキシカルアドレスに格納された変数の値を返
す。lexical-address-lookupは、変数の値が記号 *unassigned*
であればエラーを発⽣させる。46 また、指定したレキシカルアドレ
スの変数の値を変更する演算を実装する⼿続き lexical-address-
set!を書け。
練習問題 5.40: コンパイラを修正し、上で述べたコンパイル時環
境を維持するようにせよ。つまり、compileとさまざまなコード⽣
成器に compile-time-environmentという引数を追加し、compile-
lambda-bodyの中でそれを拡張するようにせよ。
練習問題 5.41: 引数として変数とコンパイル時環境を取り、その
環境に対するその変数のレキシカルアドレスを返す⼿続き find-
variable を書け。例えば、上で⽰されたプログラムの断⽚では、
式 ⟨e1⟩をコンパイルしている間のコンパイル時環境は ((y z) (a
b c d e) (x y))となる。find-variableは以下のような結果を
返す。
(find-variable 'c '((y z) (a b c d e) (x y)))
(1 2)
(find-variable 'x '((y z) (a b c d e) (x y)))
(2 0)
(find-variable 'w '((y z) (a b c d e) (x y)))
not-found

練習問題 5.42:練習問題 5.41の find-variableを使って compile-
variableと compile-assignmentを書き直し、レキシカルアドレ
ス命令を出⼒するようにせよ。find-variableが not-foundを返
す場合 (つまり、変数がコンパイル時環境内には存在しない場合)
には、以前と同じように、コード⽣成器が環境命令を使って束縛を
検索するようにする (コンパイル時に⾒つからない変数が存在し
うる位置はグローバル環境のみである。グローバル環境は実⾏環境

46これは、内部定義を取り除くための⾛査⼿法を実装するのであれば必要となる、変数
検索に対する修正です (練習問題 5.43)。レキシカルアドレッシングを動作させるために
は、内部定義を取り除く必要があります。

の⼀部だが、コンパイル時環境の⼀部ではない。47 そのため、やろ
うと思えば、評価器の演算に envにある実⾏時環境をすべて検索
させるのではなく、直接グローバル環境を⾒に⾏くようにすること
もできる。グローバル環境は演算 (op get-global-environment)
によって取得できる)。修正したコンパイラを、この節の最初に出
てきた⼊れ⼦の lambdaの組み合わせのような、いくつかの簡単な
例でテストせよ。
練習問題 5.43: 4.1.6 節では、ブロック構造の内部定義は “本物
の”defineと考えるべきでなく、⼿続きの本体は、定義されている
内部変数が set!を使って正しい値に初期化された通常の lambda
変数であるかのように解釈されるべきだと主張した。4.1.6節と練
習問題 4.16では、メタ循環インタプリタを修正して内部定義を⾛
査して削除することによってこれを達成する⽅法を⽰した。コン
パイラを修正し、⼿続きの本体をコンパイルする前にこれと同じ
変形を実⾏するようにせよ。
練習問題 5.44: この節では、レキシカルアドレスを⽣成するため
にコンパイル時環境を使うことに焦点を当てた。しかし、コンパイ
ル時環境にはそのほかの使い⽅もある。例えば、練習問題 5.38で
は基本⼿続きをオープンコード化することによってコンパイル済
みコードの効率を向上させた。我々の実装では、オープンコード
の⼿続きの名前を予約語として扱った。もしプログラムがそのよ
うな名前を再束縛しようとすると、練習問題 5.38で記述したメカ
ニズムは、新しい束縛を無視して、依然としてそれを基本⼿続き
としてオープンコード化するだろう。例えば、xと yの線形結合を
求める次の⼿続きについて考える。
(lambda (+ * a b x y)

(+ (* a x) (* b y)))

これは、例えば +matrix, *matrix、4つの⾏列といった引数で呼ぶ
ことができる。しかし、オープンコードのコンパイラは、(+ (* a

47レキシカルアドレスはグローバル環境内の変数にアクセスするためには使えません。
グローバル環境の名前はいつでも対話的に定義・再定義ができるからです。練習問題
5.43のように内部定義を取り除くと、コンパイラから⾒える定義はグローバル環境に影
響するトップレベルのものだけになります。定義をコンパイルしても、定義した名前が
コンパイル時環境に⼊ることはありません。

x) (* b y))内の +と *を依然として基本⼿続きの +と *として
オープンコード化してしまうだろう。オープンコードのコンパイ
ラを変更し、基本⼿続きの名前を含む式を正しいコードにコンパ
イルするためにコンパイル時環境を参照するようにせよ (このコ
ードは、プログラムがこれらの名前に defineや set!を⾏わない
限り正しく動作する)。

5.5.7 コンパイル済みコードと評価器の接続
ここまで、コンパイル済みコードを評価器にロードしたり、実⾏したりする⽅
法については触れてきませんでした。ここでは、明⽰制御評価器が5.4.4節の
通りに定義されており、Footnote 38で規定した追加命令も含んでいると仮定
し、この評価器に compile-and-goという⼿続きを実装します。この⼿続きは、
Scheme式をコンパイルし、その結果できるオブジェクトコードを評価器マシ
ンにロードし、評価器にそのコードを評価器のグローバル環境の中で実⾏させ、
結果を表⽰し、評価器のドライブループに⼊るというものです。また、解釈さ
れる式が、解釈される⼿続きだけでなく、コンパイル済み⼿続きも呼び出せる
ように評価器を修正します。こうすると、コンパイル済み⼿続きをマシンに⼊
れて、評価器にそれを呼び出させることができます。
(compile-and-go
'(define (factorial n)

(if (= n 1)
1
(* (factorial (- n 1)) n))))

;;; EC-Eval value:
ok
;;; EC-Eval input:
(factorial 5)
;;; EC-Eval value:
120

評価器がコンパイル済み⼿続きを扱えるようにする (例えば、上記の factorial
の呼び出しを評価する) ためには、apply-dispatch(5.4.1節)のコードを変更
して、コンパイル済み⼿続きを (複合⼿続きや基本⼿続きと別のものとして)識
別し、コンパイル済みコードのエントリポイントに制御を直接移動させる必要

があります。48

apply-dispatch
(test (op primitive-procedure?) (reg proc))
(branch (label primitive-apply))
(test (op compound-procedure?) (reg proc))
(branch (label compound-apply))
(test (op compiled-procedure?) (reg proc))
(branch (label compiled-apply))
(goto (label unknown-procedure-type))

compiled-apply
(restore continue)
(assign val (op compiled-procedure-entry) (reg proc))
(goto (reg val))

compiled-applyで continueを復元していることに注意してください。評価器
は、apply-dispatchの時点で継続がスタックの⼀番上になるようになってい
ます。⼀⽅、コンパイル済みコードのエントリポイントは、継続が continue
の中にあることを期待しています。そのため、コンパイル済みコードが実⾏さ
れる前に continueを復元する必要があります。
評価器の開始時にコンパイル済みコードを実⾏できるようにするために、

評価器の最初に branch命令を追加します。この命令は、flagレジスタが設定
されていれば、マシンを新しいエントリポイントに⾏かせるというものです。
49

(branch (label external-entry)) ; flagが設定されていれば分
岐する

48もちろん、コンパイル済み⼿続きは解釈される⼿続きと同じく複合⼿続き (⾮基本⼿
続き)です。明⽰制御評価器で使った⽤語との⼀貫性のために、この節では “複合⼿続
き”を (コンパイル済みではなく)解釈される⼿続きを指すために使います。

49評価器が branchから始まるようになったので、評価器を開始する前には常に flag
レジスタを初期化する必要があります。マシンを通常の REPLから開始させるためには、
以下のコードを使います。
(define (start-eceval)

(set! the-global-environment (setup-environment))
(set-register-contents! eceval 'flag false)
(start eceval))

read-eval-print-loop
(perform (op initialize-stack))
. . .

external-entryは、命令列の位置が valに⼊っている状態でマシンが開始さ
れると仮定します。命令列は、結果を valに⼊れるもので、末尾は (goto (reg
continue))となっています。このエントリポイントから開始すると、valに指
定された位置にジャンプしますが、実⾏が print-resultに戻るように最初に
continueを設定します。print-resultは valの値を表⽰し、それから評価器
の REPLの最初に移動します。50

external-entry
(perform (op initialize-stack))
(assign env (op get-global-environment))
(assign continue (label print-result))
(goto (reg val))

これで、次の⼿続きを使うと、⼿続き定義をコンパイルし、コンパイル済みコ
ードを実⾏し、REPLを実⾏して⼿続きを試してみることができるようになり
ます。コンパイル済みコードには、結果を valに⼊れて continueの位置に戻
ってきてほしいので、式はターゲット valとリンク returnでコンパイルしま
す。コンパイラによって⽣成されるオブジェクトコードを評価器レジスタマシ
ンで実⾏可能な命令に変換するために、レジスタマシンシミュレータ (5.2.2節)
の⼿続き assembleを使います。それから valレジスタを命令リストを指すよ

50コンパイル済み⼿続きはシステムが表⽰を試みるかもしれないオブジェクトなので、
システムの表⽰演算 user-print(4.1.4節)も修正し、コンパイル済み⼿続きの部品を表
⽰させないようにします。
(define (user-print object)

(cond ((compound-procedure? object)
(display (list 'compound-procedure

(procedure-parameters object)
(procedure-body object)
'<procedure-env>)))

((compiled-procedure? object)
(display '<compiled-procedure>))

(else (display object))))

うに初期化し、flagを評価器が external-entryに⾏くように設定して、評価
器を開始します。
(define (compile-and-go expression)

(let ((instructions
(assemble
(statements
(compile expression 'val 'return))

eceval)))
(set! the-global-environment (setup-environment))
(set-register-contents! eceval 'val instructions)
(set-register-contents! eceval 'flag true)
(start eceval)))

5.4.4節の終わりでやったようにスタック監視を設定すると、コンパイル済み
コードのスタック使⽤を調査できます。
(compile-and-go
'(define (factorial n)

(if (= n 1)
1
(* (factorial (- n 1)) n))))

(total-pushes = 0 maximum-depth = 0)
;;; EC-Eval value:
ok
;;; EC-Eval input:
(factorial 5)
(total-pushes = 31 maximum-depth = 14)
;;; EC-Eval value:
120

この例と、5.4.4節の終わりに⽰した同じ⼿続きの解釈版を使った (factorial
5) の評価を⽐べてみてください。解釈版は、プッシュ 144回と、最⼤スタック
深度 28を必要としていました。これは、コンパイル戦略による最適化のよい
例となっています。

解釈とコンパイル
この節のプログラムを使うと、解釈とコンパイルという実⾏戦略の選択肢を実
験できるようになります。51 インタプリタはマシンをユーザープログラムのレ
ベルに上げるもので、コンパイラはユーザープログラムを機械語のレベルに落
とすものです。Scheme⾔語は (Scheme⾔語だけでなく、どんなプログラミン
グ⾔語でも)、機械語の上に建てられた⼀貫性のある抽象化の⼀族と⾒なすこ
とができます。インタプリタは、プログラム実⾏のステップがこれらの抽象化
によって構成されていて、そのことによってプログラマに理解しやすくなって
いるため、対話的プログラム開発やデバッグに向いています。コンパイル済み
コードは、プログラム実⾏のステップが機械語によって構成されており、また
コンパイラは⾼レベルの抽象化の壁を越えて最適化を⾃由に⾏えるため、実⾏
速度が速くなります。52

解釈とコンパイルという⼆つの選択肢は、⾔語を新しいコンピュータに移
植する際にもそれぞれ異なった戦略につながります。例えば、Lispを新しいマ
シンに実装したいとしましょう。ひとつの戦略は、5.4節の明⽰制御評価器を
ベースにして、その命令を新しいマシンの命令に翻訳するというものです。も
うひとつの戦略は、コンパイラをベースにして、新しいマシンのコードを⽣成
するようにコード⽣成器を作り替えるというものです。⼆つ⽬の戦略を取る場

51コンパイラを拡張して、コンパイルされたコードが解釈される⼿続きを呼び出せる
ようにすると、さらによくなります。練習問題 5.47参照。

52ユーザープログラムの実⾏時にエラーにぶつかったとき、システムを⽌めたり間違
った答えを出したりすることを許すのではなく、エラーを検出してシグナルを発⽣させ
ることを求めると、実⾏戦略を問わず、かなりのオーバーヘッドを引き起こすことにな
ります。例えば、範囲外の配列参照は、実⾏前に参照の有効性をチェックすることによっ
て検出できます。しかし、チェックのオーバーヘッドは配列参照そのもののコストの何
倍にもなる可能性があるので、そのようなチェックが望ましいのかどうかを決める際に
はプログラマは速度と安全性をはかりにかける必要があります。良いコンパイラは、そ
のようなチェックのあるコードを⽣成する能⼒を持ち、冗⻑なチェックを避け、コンパ
イル済みコード中のエラーチェックの程度や種類をプログラマが制御できるようにする
べきです。

Cや C++のような⼈気のある⾔語のコンパイラは、実⾏をできるだけ⾼速にするた
めに、エラーチェック演算を実⾏コードにほとんど⼊れません。その結果、明⽰的にエラ
ーチェックを⾏うことがプログラマに求められます。しかし、残念ながら、スピードが制
約とならない重要なアプリケーションですら、そのことがおろそかにされることがよく
あります。そのように書かれたプログラムは、速いと同時に危険なものになります。例え
ば、1988年にインターネットを⿇痺させた悪名⾼い “ワーム”は、fingerデーモンで⼊⼒
バッファがオーバーフローしたかどうかというチェックがされていないという unix(tm)
オペレーティングシステムの⽋陥を利⽤していました (Spafford 1989参照)。

合、プログラムをまず元の Lispシステムで動くコンパイラでコンパイルして、
それをコンパイル済みの実⾏時ライブラリとリンクすることで、どんな Lispプ
ログラムでも実⾏できるようになります。53 それより良いのは、コンパイラ⾃
⾝をコンパイルして、それを新しいマシンで動かしてほかの Lispプログラムを
コンパイルするということです。54 また、4.1節のインタプリタのひとつをコ
ンパイルして、新しいマシンで動くインタプリタを作るということも可能です。

練習問題 5.45: コンパイル済みコードと評価器が同じ計算を⾏う
際に使うスタック演算を⽐較することで、コンパイラがスタック
使⽤の最適化をどの程度⾏っているかを、速度 (スタック演算の総
数を減らすこと)と空間 (スタックの最⼤深度を減らすこと)の両
⾯から判断できる。この最適化されたスタック使⽤と、それと同
じ計算をするための特別⽬的マシンの性能を⽐較すると、コンパ
イラの品質についてのある種の指標が得られる。

a. 練習問題 5.27では、上記の再帰階乗⼿続きを使って n!を計
算するのに評価器が必要とするプッシュの数と最⼤スタック
深度を nの関数として表すよう求めた。練習問題 5.14では、
図 5.11に⽰した特別⽬的階乗マシンについて、同じ測定を⾏
うよう求めた。ここでは、コンパイルした factorial⼿続き
について同じ分析を⾏う。
コンパイル版のプッシュの数と解釈版のプッシュの数の⽐を
求めよ。また、スタック深度についても同じことを⾏え。n!
を計算するのに使われる演算数とスタック深度は nに線形な
ので、nが⼤きくなるにつれてこれらの⽐は定数に近づくは

53もちろん、解釈とコンパイルのどちらの戦略を取る場合でも、メモリ割り当て、⼊出
⼒その他のこれまで評価器やコンパイラの議論の中で “基本的”としてきたさまざまな演
算も実装する必要があります。ここで、できるだけ作業量を減らすひとつの戦略として、
これらの演算を可能な限り Lispで書いて、それらを新しいマシン⽤にコンパイルすると
いうものがあります。究極的には、新しいマシンのために⼿書きするものは、(ガベージ
コレクションや実際のマシンの基本⼿続きを適⽤するメカニズムなどの)⼩さなカーネ
ルにすべてが集約されることになります。

54この戦略を取る場合、コンパイルしたコンパイラを新しいマシンで動かしたものに
よってプログラムをコンパイルした結果が元の Lispシステムでコンパイルしたものと同
⼀になるかといった、コンパイラの正確性についての⾯⽩いテストをすることになりま
す。違いの原因を突き⽌めるのは楽しい作業なのですが、イライラすることもよくあり
ます。結果はほんのわずかな詳細に⼤きく影響されるからです。

ずだ。それらの定数を答えよ。同様に、特別⽬的マシンのス
タック使⽤と解釈版のスタック使⽤の⽐を求めよ。
特定⽬的マシンのコードに対する解釈されるコードの⽐率
と、コンパイル済みコードに対する解釈されるコードの⽐率
を⽐較せよ。特定⽬的マシンがコンパイル済みコードよりも
ずっと性能がいいことがわかるだろう。⼿書きによるコント
ローラのコードは、我々の初歩的な汎⽤⽬的コンパイラに⽣
成されるものよりずっと優れているからだ。

b. ⼿書き版により近い性能のコードを⽣成するようコンパイラ
を改良するアイデアを提案できるか。

練習問題 5.46: ⽊の再帰による以下のフィボナッチ⼿続きにつ
いて、
(define (fib n)

(if (< n 2)
n
(+ (fib (- n 1))

(fib (- n 2)))))

これをコンパイルしたときの性能と、図 5.12の特別⽬的フィボナ
ッチマシンを使う場合の性能とを⽐較し、練習問題 5.45のような
分析を実⾏せよ (解釈版のパフォーマンス測定については 練習問
題 5.29参照)。フィボナッチ数の計算では、使われる時間的リソー
スは nに対して線形ではないため、スタック演算の⽐は nと独⽴
な極限値には近づかない。
練習問題 5.47: この節では、解釈されるコードがコンパイル済み
⼿続きを呼び出せるように明⽰制御評価器を修正する⽅法につい
て記述した。コンパイル済み⼿続きが、基本⼿続きとコンパイル
済み⼿続きだけでなく、解釈される⼿続きも呼べるようにコンパ
イラを修正する⽅法を⽰せ。このためには、compile-procedure-
call を修正し、(解釈される)複合⼿続きの場合を扱うようにする
必要がある。compile-proc-appl と同じ target と linkage の組
み合わせをすべて扱うよう気をつけること。実際の⼿続き適⽤を
⾏うには、コードは評価器の compound-applyエントリポイントに
ジャンプしなければならない。このラベルはオブジェクトコード
からは直接参照できない (アセンブル中のコードによって参照さ

れるラベルはすべてそのコード内で定義したものであるようアセ
ンブラが要求することによる)。そのため、compapp というレジス
タを評価器マシンに追加してそのエントリポイントを持たせるよ
うにして、このレジスタを初期化する命令を追加する。
(assign compapp (label compound-apply))
(branch (label external-entry)) ; flag が設定されて
いれば分岐する
read-eval-print-loop . . .

コードをテストするには、まず⼿続き gを呼ぶ⼿続き fを定義す
る。それから compile-and-goを使って fの定義をコンパイルし、
評価器をスタートさせる。そこで、評価器に対し⼊⼒を⾏って g
を定義し、fの呼び出しを試みる。
練習問題 5.48: この節で実装した compile-and-goインターフェ
イスは、⼀回 (評価器開始時)しか呼ぶことができないため、扱い
にくい。明⽰制御評価器から次のようにしてコンパイラを呼び出
すことができる compile-and-run という基本⼿続きを追加して、
コンパイラ-インタプリタ間のインターフェイスを拡張せよ。
;;; EC-Eval input:
(compile-and-run
'(define (factorial n)

(if (= n 1) 1 (* (factorial (- n 1)) n))))
;;; EC-Eval value:
ok
;;; EC-Eval input:
(factorial 5)
;;; EC-Eval value:
120

練習問題 5.49:明⽰制御評価器の REPLの代替⼿段として、REPL
を⾏うレジスタマシンを設計せよ。つまり、マシンは式を読み込
み、それをコンパイルし、その結果のコードをアセンブルして実
⾏し、結果を出⼒するというループを⾛らせるということになる。
これは、シミュレーション構成の中では簡単に実⾏できる。⼿続
き compileと assembleを “レジスタマシン演算”として呼ぶよう
にすればよいからだ。

練習問題 5.50: 4.1節のメタ循環評価器をコンパイラを使ってコン
パイルし、レジスタマシンシミュレータを使ってそのプログラム
を実⾏せよ (⼀度に複数の定義をコンパイルするには、それらの定
義を beginの中にまとめればよい)。その結果できるインタプリタ
は、複数レベルで解釈が⾏われるため、実⾏速度はとても遅くな
るだろう。しかし、そのすべての詳細をうまく動かせるようにす
ることは学びの多い練習となる。
練習問題 5.51: 5.4節の明⽰制御評価器を翻訳することによって、
C⾔語 (または好きな低レベル⾔語)による Schemeの基本的な実
装を開発せよ。このコードを動かすためには、適切なメモリ割り
当てルーチンや、その他の実⾏時サポートを⽤意する必要がある
だろう。
練習問題 5.52: 練習問題 5.51と対応するものとして、コンパイラ
を修正して Schemeの⼿続きを C⾔語の命令列にコンパイルする
ようにせよ。4.1節のメタ循環評価器をコンパイルして、C⾔語で
書かれた Schemeインタプリタを⽣成せよ。

索引

この索引に間違いがあるとしたら、コンピュータの助けを
借りて作ったことによるものだろう。
---Donald E. Knuth, Fundamental Algorithms
(Volume 1 of The Art of Computer Programming)

k-term finite continued fraction, 75
k項有限連分数, 75

acquire, 339

applicative-order evaluation, 17

arbiter, 342

average damping, 74

bound variable, 29

Church-Turing thesis, 420

closure property, 104
code generator, 622
coerce, 214
coercion, 209
combination, 6

compilation, 619
compile-time environment, 660

compound data, 85
compound data object, 85
compound procedure, 12
computability, 420
computational process, 1
concurrently, 323
congruent modulo, 53
connector, 311
consequent expression, 18

constraint network, 311

continuation procedure, 465

control structure, 505
controller, 538
conventional interface, 87, 121
current time, 307

data, 1, 96
data abstraction, 86, 88
data path, 538
data-directed, 183
data-directed programming, 88, 193
deadlock, 343
deadlock-recovery, 343
debug, 2
deep binding, 414
deferred operation, 35
delayed argument, 380
delayed evaluation, 234, 346
delayed object, 349

dependency-directed backtracking, 453
depth-first search, 453
deque, 289
derived expression, 407
digital signal, 297
dispatching on type, 192

driver loop, 417

enclosing environment, 254
entry point, 541
enumerator, 123
environment, 9, 254
environment model, 234
Euclid's Algorithm, 50

evaluation, 6
evaluator, 394
event-driven simulation, 297
evlis tail recursion, 603
execution procedure, 429
explicit-control evaluator, 597
expression, 6

failure continuation, 465
FIFO, 284
filter, 65, 123
first-class, 80
fixed point, 72
fixed-length, 174
forcing, 437
forwarding address, 592
frame, 254, 495
frame coordinate map, 144
frame number, 659
framed-stack, 601
free, 29
free list, 587

full-adder, 299
function box, 297
functional programming, 247
functional programming language, 389

garbage, 590
garbage collection, 583, 590
garbage collector, 272
garbage-collect, 441
generic operation, 88
generic procedure, 178, 183
glitch, 2
global, 32, 254
global environment, 9
golden ratio, 39
grammar, 459

half-interval method, 70

Halting Theorem, 423
headed list, 289
hiding principle, 238
hierarchical, 105
hierarchy of types, 212
higher-order procedure, 59
Horner's rule, 128

imperative programming, 252
indeterminate, 218
index, 584
indexing, 496
instantiated with, 485
instruction, 537, 541
instruction counting, 581
instruction execution procedure, 564
instruction sequence, 624
instruction tracing, 581
integer, 5
integerizing factor, 230
integrator, 375
interning, 586
interpreter, 3, 394
invariant quantity, 47
inverter, 297
iterative improvement, 83
iterative process, 35

key, 172

label, 541
lazy evaluation, 435
lexical address, 659
lexical addressing, 414
lexical scoping, 31
linear iterative process, 35
linear recursive process, 35
linkage descriptor, 623
list, 106, 111
list structure, 106
list-structured, 91
list-structured memory, 582

local evolution, 32
local state variable, 235
location, 583
logic-programming, 395
logical and, 297
logical deduction, 492
logical or, 297

machine language, 619
macro, 407
map, 123
mark-sweep, 591
memoization, 42, 295
memoize, 437
merge, 390
message passing, 98, 200
message-passing, 240
metacircular, 396
Metalinguistic abstraction, 393
Miller-Rabin test, 58
modular, 233
modulo, 53
modus ponens, 504
moments in time, 323
Monte Carlo integration, 245
Monte Carlo simulation, 243
mutable data object, 271
mutator, 271
mutex, 339
mutual exclusion, 339

native language, 619
needed, 625
network, 393
Newton's method, 77
nil, 107
non-computable, 423
non-strict, 435
nondeterministic, 329
nondeterministic choice point, 452
nondeterministic computing, 395, 449
normal-order, 435

normal-order evaluation, 17, 395

obarray, 585

object program, 619
open-code, 657
operand, 7
operator, 7, 428
or-gate, 297
order of growth, 43

ORゲート, 297
output prompt, 417

package, 194
painter, 136
pair, 90, 91
parse, 459
Pascal's triangle, 43
pattern, 484
pattern matcher, 495
pattern matching, 494
pattern variable, 484
pipelining, 324
pointer, 103
poly, 219
power series, 363
predicate, 18
prefix, 174
prefix code, 174
prefix notation, 7
pretty-printing, 7
primitive constraint, 311
probabilistic algorithm, 55
procedural abstraction, 27
procedural epistemology, xxv
procedure, 4, 36
procedure definition, 12
process, 36
program, 1
programming language, 2
prompt, 417

pseudo-random, 243
pseudodivision, 230
pseudoremainder, 230

quasiquote, 627
query, 480
query language, 480
queue, 283
quote, 153

Ramanujan number, 374
rational function, 227
RC circuit, 375
RC回路, 375
read-eval-print loop, 8
reader macro character, 529
real number, 5
rear, 283
recursion equation, 3
Recursion theory, 421
recursive, 10, 26
recursive process, 35
red-black tree, 169

register, 537
register machine, 537
register table, 564
release, 339
REPL, 8
resolution principle, 478
ripple-carry adder, 302
robust, 152
RSA algorithm, 55
RSAアルゴリズム, 55
rule, 481, 488

scope, 29
selector, 88
semaphore, 339
separator code, 174
sequence, 64, 105

serializer, 330, 332
series RLC circuit, 382
shadow, 255
shared, 278
side-effect bug, 251
sieve of Eratosthenes, 356
smoothing, 82
source language, 619
source program, 619
sparse, 223

stack, 36, 555
state variable, 35, 235
statement, 625
stop-and-copy, 590
stratified design, 151
stream, 234, 345, 346
stream processing, 17
strict, 435
subroutine, 550
substitution, 16
substitution model, 15
subtype, 212

summer, 375
supertype, 212
symbolic expression, 87
syntactic sugar, 12

system, 393
systematically search, 452

tabulation, 42, 295

tail-recursive, 36, 607
target, 623

thunk, 437
time, 323

time segment, 307
tower, 212
tree, 115
tree accumulation, 10
tree recursion, 38
truth maintenance, 453
Turing machine, 420
type field, 585
type tag, 183, 188
type-inferencing, 385
typed pointer, 584

unbound, 255
unification, 478, 494, 500
unification algorithm, 478
univariate polynomial, 218
universal machine, 419
upward-compatible extension, 445

value, 8, 153
value of a variable, 254
variable, 8
variable-length, 174
vector, 583

width, 101
wire, 297
wishful thinking, 89

zero crossing, 376

アキュムレータ, 241
アクション, 545
アセンブラ, 562
アトミック, 341
アドレス, 583
アドレス演算, 583
アービタ, 342
イベント駆動シミュレーション, 297
インタプリタ, 3, 394
インデクシング, 496
インデックス, 584

インバータ, 297
エイリアス, 251
エブリス末尾再帰, 603
エラトステネスのふるい, 356
エントリポイント, 541
オブアレイ, 585
オブジェクト, 234
オブジェクトプログラム, 619
オープンコード, 657
カプセル化, 238
カーマイケル数, 55
ガベージコレクション, 441, 583, 590
ガベージコレクタ, 272
キャッシュコヒーレンス, 327
キャプチャ, 29
キュー, 283
キー, 172
クエリ, 480
クエリ⾔語, 480
クォート, 153
クロージャ, 87
グリッチ, 2
グローバル, 32, 254
グローバル環境, 9
コネクタ, 311
コメント, 133
コンストラクタ, 88
コントローラ, 538
コンパイル, 619
コンパイル時環境, 660
コンビネータ, 428
コード⽣成器, 622
ゴミ, 590
サブタイプ, 212
サブルーチン, 550
サンク, 437
システム, 393
シリアライザ, 332
シンタックスシュガー, 12
ジェネリック⼿続き, 178, 183
ジェネリック演算, 88
スコープ, 29

スタック, 36, 555
スタックフレーム, 601
ストップ &コピー, 590
ストリーム, 234, 345, 346
ストリーム処理, 17
スラッシング, ix
スーパータイプ, 212
セマフォ, 339
セル, 340
セレクタ, 88
ゼロ交差, 376
ソースプログラム, 619
ソース⾔語, 619
タイプタグ, 183, 188
タグアーキテクチャ, 585
タブロー, 368
タワー, 212
ターゲット, 623
チャーチ・チューリングのテーゼ, 420
チャーチ数, 99
チューリングマシン, 420
テーブル化, 42, 295
デジタル信号, 297
デッドロック, 343
デッドロック回復, 343
デバッグ, 2
データ, 1, 96
データパス, 538
データ主導, 183
データ主導プログラミング, 88, 193
データ抽象化, 86, 88
ドット末尾記法, 111
ドライバループ, 417
ニュートン法, 77
ネイティブ⾔語, 619
ネットワーク, 393
バグ, 2
バッククォート, 627
バックトラック, 453
バランスが取れている, 119
バランスの取れた, 169
バリア同期, 344

パイプライン, 324
パスカルの三⾓形, 43
パターン, 484
パターンマッチャ, 495
パターンマッチング, 494
パターン変数, 484
パッケージ, 194
ファーストクラス, 80
フィルタ, 65, 123
フレーム, 254, 495
フレーム数, 659
ブレークポイント, 581
ブロック構造, 31
プリティプリント, 7
プログラミング⾔語, 2
プログラム, 1
プロセス, 36
プロンプト, 417
ベクタ, 583
ベルトランの仮説, 360
ベースアドレス, 584
ペア, 90, 91
ペインタ, 136
ホーナー法, 128
ポインタ, 103
マクロ, 407
マップ, 123
マーク &スイープ, 591
マージ, 390
ミューテックス, 339
ミューテータ, 271
ミラー-ラビンテスト, 58
メタ循環, 396
メタ⾔語抽象化, 393
メッセージパッシング, 98, 200, 240
メモ化, 42, 295, 437
モジュール式, 233
モンテカルロシミュレーション, 243
モンテカルロ積分, 245
モーダス・ポネンス, 504
ユニフィケーション, 478, 494, 500
ユニフィケーションアルゴリズム, 478

ユークリッドの互除法, 50
ユークリッド環, 228
ラベル, 541
ラマヌジャン数, 374
リスト, 106, 111
リスト構造, 91, 106
リスト構造メモリ, 582
リンク記述⼦, 623
リーダーマクロ⽂字, 529
レキシカルアドレス, 659
レキシカルアドレッシング, 414
レキシカルスコーピング, 31
レジスタ, 537
レジスタテーブル, 564
レジスタマシン, 537

⼀元多項式, 218
万能機械, 419
上位互換性のある拡張, 445
不動点, 72
不変量, 47
不定元, 218
両端キュー, 289
並⾏, 323
予定表, 304
⼆項係数, 43
代⼊演算⼦, 235
代数的仕様, 97
位置, 583
体系的に探索, 452
依存主導バックトラック, 453

値, 8, 153
停⽌性問題, 422
先⼊れ先出し, 284
先端, 283
全加算器, 299
共有, 278
具体化される, 485
再帰プロセス, 35
再帰⽅程式, 3
再帰理論, 421

再帰的, 10, 26
冪級数, 363
出⼒プロンプト, 417
分離符号, 174
列, 64, 105
列加速, 367
列挙, 123
制御構造, 505
制約ネットワーク, 311
前置記法, 7
副作⽤バグ, 251
加法的, 88, 182, 192
加算器, 375
区間⼆分法, 70
半加算器, 298
厳密, 435
参照透明, 250
反復プロセス, 35
反復改良法, 83
可変データオブジェクト, 271
可変⻑, 174
合成, 81
名前呼び, 354, 437
名前呼びサンク, 354
命令, 537, 541
命令トレース, 581
命令列, 624
命令型プログラミング, 252
命令実⾏⼿続き, 564
命令数カウンタ, 581
命令⽂, 625
固定⻑, 174
圧縮, 591
型つきポインタ, 584
型によるディスパッチ, 192
型の階層, 212
型フィールド, 585
型推論, 385
基本制約, 311
場合分け, 17
増加オーダー, 43
変位数, 659

多倍⻑整数, 585

失敗継続, 465
実数, 5
実⾏⼿続き, 429

導出原理, 478

局所状態変数, 235
希望的思考, 89

平滑化, 82

引数, 7
強制, 437

必要とする, 625
必要呼び, 354, 438

成功継続, 465

⽊の再帰, 38
⽊の集積, 10
未束縛, 255

構⽂, 397
構⽂解析, 459

正規順序, 435

奥付

表紙に描かれているのは、Agostino Ramelliによって 1588年に作られたブック
ホイール機構です。これは、初期のハイパーテキストナビゲーション⽀援とい
えるでしょう。この版画の画像は New Gottlandの J. E. Johnsonによってホスト
されています。
本⽂のフォントは Linux Libertineで、⾒出しは Linux Biolinumです。どちら

も Philipp H. Pollによるものです。等幅フォントは、Raph Levienによって作ら
れた Inconsolataを Dimosthenis Kaponisと Takashi Tanigawaが Inconsolata lgcと
いう形で補完したものを使っています。訳注 7 表紙のフォントは Juan Pablo del
Peralによってデザインされた Alegreyaです。
グラフィックデザインとタイポグラフィは Andres Raba の⼿によります。

Texinfoのソースは Perlスクリプトで LaTeXに変換され、XeLaTeXにより pdf
にコンパイルされています。図は Inkscapeで描かれたものです。

訳注 7⽇本語版では源ノ⾓ゴシック・源ノ明朝を使っています。

